
Capítulo  I.  INTRODUCCIÓN A LAS LIBRERÍAS API’S 32 DE WINDOWS.

Analicemos antes que nada ciertas cositas, por ejemplo, “LIBRERÍA”, si vamos a una librería (y hablo de un establecimiento que vende puros libros), lo primero que ustedes ven es que los libros están ordenados por temas, por ejemplo, si quieres libros de leyes, pues irán a la sección de leyes, si quieres idiomas irán a la sección de idiomas y así sucesivamente, después que llegas a la sección que quieres ¿que ves?. Pues libros!, obvio no!. ¿Pero ves los mismos libros? ¡No verdad! Veras diferentes libros que tratan diferentes temas sobre la sección escogida, es decir, cada libro tiene un fin en común o un propósito especifico. Que tiene que ver libros con API Win32. 

Sencillo, imaginemos que nuestra librería es Windows, en nuestra librería encontramos departamentos kernel.dll, guid32.dll, etc, en nuestro departamento encontramos los libros que son las funciones API Win32 (Interfaz de programación de aplicaciones),  así de simple!. Cada API no es nada mas que una simple función que tiene un propósito único y uniéndolas o interactuando con ellas se pueden hacer cosas realmente útiles y sorprendentes. Ahora yo dije que la librería era Windows, bueno esto lo hice para que vieran el ejemplo, en realidad se llama librerías a las dll, pero creo que como lo explique se ve mas claro.

Pero ustedes dirán ¿Cómo accedemos a las API’S? Pues este manual les enseñara, como se usan y como accedemos a ella, vamos a usar el lenguaje Visual Basic 6.0 para ello, después que conozcan y entiendan las API’S pueden llevarlo a cualquier lenguaje como Visual C++, VJ++, Visual Foxpro, etc. Inclusive el aprender el lenguaje VC++, se hace mucho mas fácil si conocen las API’S

COSAS DE WINDOWS

Todos nos sorprendimos cuando saltamos de Windows 3.1 a Windows 95, se podría decir que a mi manera de ver ese fue el salto mas grande, ya sea en interfaz, programación, facilidad, etc, ya que XP para mi es los mismo pero mas bonito. Alguna vez nos hemos preguntado como programadores, ¿como funciona todo esto? , en realidad no podemos saberlo ya que los códigos de Windows permanecen en secreto en Microsoft, pero viendo su interfaz y su funcionamiento mas o menos podemos deducir como Windows hace las cosas, por ejemplo, imaginemos que tenemos 4 ventanas abiertas y que las vemos al mismo tiempo, ustedes le dan clic a una de las ventanas y ella se marca y si estaba en el fondo pues se pone de primera, nadie se ha preguntado ¿Cómo sabe Windows que esa ventana estaba ahí?, y no solo eso, imaginen que tienen Excel y Word abierto y le dan a Word y luego entran al menú de Word, ¿Cómo sabe Windows que le diste a Word y estas abriendo el menú de Word y no de Excel?. Es ahí donde entra esta palabrita:

HWND (IDENTIFICADOR DE VENTANAS).

El hwnd es un número, ahora ¿que función cumple ese número?, imaginemos que tenemos 3 aplicaciones abiertas Word, Excel, y PowerPoint, y las tenemos abierta al mismo tiempo, cuando damos clic a una de ellas en la pantalla,  Windows sabe a que programa se da clic gracias al HWND, cada ventana (ojo ventana no programa, ya explicare esto con mas detalle), posee un número único al momento de ejecutarse, este número único entra en un registro de ventanas abiertas en el  momento de ejecutarlo, cuando se le da clic a una ventana o cualquier otro mensaje él (Windows) sabe para quien va el mensaje ya que el tiene un registro de las ventanas abiertas, y mediante un algoritmo ejecuta la función encargada de controlar el mensaje.

Que quise decir con “OJO ventana no programa”, todo lo que tu ves en Windows es una ventana, en este momento te acabas de sorprender con lo que dije pero es lo correcto, ABSOLUTAMENTE TODO lo que estas viendo es una ventana, desde un botón, un listbox, un combobox, un menú, TODO!, ¿que significa? que si tu tienes 100 botones, cuando tu aplicación arranca se generan 100 HWND únicos para cada componente, es obvio ver de que tienen que ser único ya que si se repitiera alguno por error, eso podría crear un colapso en Windows ya que el no sabría a quien enviarle el mensaje. Este número hwnd no lo colocas tu, ni puedes colocárselo, Windows es el encargado de colocar ese número cuando una aplicación arranca, y se lo coloca a todo lo que sea ventana, de esta manera es obvio pensar que Windows controla los clic, y todos lo demás mensajes por este número de identificación. Para que lo vean mas fácil, es como una biblioteca la cota de los libros serian en este caso una especie de hwnd, ya que cada libro posee una cota única y no se puede repetir, el bibliotecario por el número sabe la ubicación del libro y procede a buscarlo, parecido sucede en Windows, de alguna manera Windows debe de controlar todo lo que es posicionamiento, y tratamiento de mensajes y sabe a quien va, gracias a ese número que se llama hwnd.

Es difícil ver que todo es una ventana, pero es fácil ¿Qué es un botón?. Bueno a un formulario vació, quítenle el caption (parte superior donde esta el texto y los botones de maximizar, cerrar y minimizar), déjenle los bordes, póngalo gris, y coloquen un texto en el centro y que tienen ¡UN BOTON!, ¿qué es un combobox?. Un formulario con los bordes, sin el caption, color de fondo blanco y un botón al lado. Difícil de imaginar!! y todo se origino con un formulario o ventana, el problema de entender esto, es que nosotros entendemos como ventana a un formulario típico y en realidad en Windows ventana es algo un poco mas abstracto, pero no se preocupen en el camino iremos aprendiendo esos detalles.

HDC (DISPOSITIVO DE CONTEXTO).

Imaginen un pintor, el pintor para pintar ¿que necesita?, bueno de manera básica sería un pincel, un cabestrillo, colores y un lienzo; el cabestrillo es mi aplicación y ella se identifica con el HWND, y el lienzo es mi HDC, es decir, el hdc es el lugar donde se pinta tan simple como eso, toda ventana (me refiero a TODO!). posee un hdc, el hdc es donde nosotros ponemos nuestra creatividad, este hdc también es un número único, que no se duplica, y solo lo usamos cuando dibujamos. Cuando veamos las API’S graficas entenderán un poco mas esto.

MENSAJERIA WINDOWS

Es fácil ver, que Windows es un lenguaje orientado a objetos o que por lo menos esa es la filosofía que quiere vender, ahora una de las cosas mas difíciles para los programadores cuando están aprendiendo todo esto de programación orientada a objetos es como hacerlos funcionales, es decir, como enviar un mensaje de un objeto a otro; ya dije que todo era una ventana, y desde un punto de vista todo es un objeto, ahora esos objetos hablan entre si, inclusive para no complicarnos, el movimiento del ratón genera un mensaje que el usuario le transmite a Windows, si ustedes se movieran desde el píxel 0 hasta el píxel 800 (ancho), increíblemente Windows esta recibiendo 800 mensajes del usuario ya que para cada movimiento en píxel se genera dicho mensaje, ahora que hace Windows, el recibe el mensaje verifica en su lista de ventanas activas “hwnd”, y verifica cual hwnd se encuentra en la posición del mouse, y el hwnd seleccionado será el ganador de dicho mensaje, si el objeto o ventana tiene un código para procesar dicho mensaje este se ejecuta en caso contrario este se omite, es decir, el papel de Windows es como el de una compañía de correo, agarra el mensaje del EMISOR y se lo envía al RECEPTOR. Es increíble pero cuando veamos mas profundo esto se van a impresionar de cómo lo creadores de Windows a  mi manera de ver hicieron un excelente y practico trabajo con este tema de mensajes, TODO absolutamente TODO gira en los mensajes.

¿COMO NOS COMUNICAMOS?

Si conocemos a una persona y la vemos en la calle inmediatamente tu procedes a saludarla en el saludo tu usas un componente de tu cuerpo para comunicarte (la boca), ese componente envía una onda (mensaje), la cual es recibida por el receptor por otro componente (oreja), y su cerebro inmediatamente reacciona al mensaje y de ahí se puede generar una conversación, es fácil ver esto, ya que lo hacemos a diario, pero ustedes dirán como hablo con Windows o como Windows habla conmigo, evidentemente Windows no habla con los componentes desde un punto de vista literario, sino desde un punto de vista informático. 

Un mensaje de Windows tiene básicamente esta estructura.

Mensaje “Componente que recibe el mensaje”, “Tipo de mensaje”, mensaje1, mensaje_opcional
Esta es una estructura básica o simple de un mensaje de Windows, mas adelante veremos la función API encargada de ello. Pero se preguntaran que significan esos 4 componentes imaginen que Windows hablara, y cuando entran a Windows el los saluda,  haría lo siguiente:

Mensaje “TU”, “SALUDOS”, “Buenos Días”, NULL.

Que significa esto: que el receptor eres “tu”, Windows te va  saludar “SALUDOS”, y te dirá “Buenos Días”, el mensaje opcional es NULL (Nada). Ahora esto no es un mensaje real de Windows es un ejemplo, pero veamos los componentes:

Mensaje = Windows usa la función API SendMessage y PostMessage para ese trabajo.

Componente que recibe el mensaje = sería el hwnd del receptor, es decir, quien va recibir el mensaje.

Tipo de mensaje = muy importante este parámetro, aquí se le pasa un número, que quiero decir, Windows tiene una lista de mensajes que los identifica por un número. Ejemplo.

1 = Movimiento del mouse

2 = Clic del mouse

3 = Tecla presionada

etc.

Estos números no son los números reales, sino estoy poniendo un ejemplo, veremos las constantes reales cuando veamos la función.

mensaje1 = mensaje que enviamos o recibimos.

mensaje_opcional = mensaje que enviamos o recibimos.

De seguro están confundidos con estos dos últimos parámetros ya que coloque “mensaje que enviamos o recibimos”, se confundirán con la palabra “recibimos”, algunas veces estas funciones no sirven solo para enviar sino también para recibir un dato u información, lo veremos claro mas adelante.

Capítulo  II.  NUESTRO PRIMER CÓDIGO Y SENDMESSAGE

3 paginas de habladuría y de seguro están ansiosos de poner sus manos en el teclado y escribir código, antes que nada voy a disculparme por que les dije que los códigos serian en VB (Visual Basic), pero nuestro primer código es en VC++. Este es el único se los prometo, todos los demás serán en VB. Algunos de ustedes se asustaran ya que la palabra C y  C++ asustan a muchos programadores, pero no tienen que conocer de ello, ya que lo que vamos analizar es el código y no la estructura del lenguaje.

Primeramente entren en VB y como VB es mágico les coloca un formulario listo para programar, si le dan F5 o play, arranca el formulario y LISTO!!, su primer programa en Windows sin haber programado nada. Pero no tienen curiosidad de saber que hay detrás de ese formulario, no quieren saber como se origina, que hay detrás de “cámara”. Pues aquí les escribo lo que hay detrás de cámara.

Pasos para hacer nuestro programa:

1. Abrir el programa VC++

2. Selecciona el menú “File” y la opción “New”

3. Busca en la solapa de “Project” la opción que dice “Win32 Application”

4. Coloca un nombre en el textbox de “Project Name” y señala una ubicación donde estará el proyecto y presionan OK.

5. En el formulario siguiente seleccionan “And empty project” y le dan finish y le dan OK al formulario que les aparece luego.

6. Abajo verán dos solapas o fichas cuyos nombres son ClassView y FileView seleccionan FileView y expanden la opción del árbol.

7. En la carpeta Source File dan al botón secundario del mouse y seleccionan “Add files to fólder...”

8. Escriben el nombre del archivo y pónganle al final la extensión .c, es decir, nombre.c y le dan SI al formulario que les aparece luego.

9. Dan doble clic a Source File y le dan doble clic al archivo que crearon, les preguntara se desean crearlo y le dicen que Si.

10.  Y luego pegan este código:

#include <windows.h>

LONG proc;

HWND hWnd;

LRESULT CALLBACK GestorMensajes(HANDLE hWnd, UINT mensaje, WPARAM wParam, LPARAM lParam)

{


switch (mensaje) {


case WM_DESTROY:



PostQuitMessage(0);



break;


default:



return DefWindowProc(hWnd,mensaje,wParam,lParam);


}

}

BOOL IniciaAplicacion(HANDLE hInstance)

{


WNDCLASS wc;


wc.style = CS_HREDRAW | CS_VREDRAW ;


wc.lpfnWndProc = GestorMensajes;


wc.cbClsExtra = 0;


wc.cbWndExtra = 0;


wc.hInstance = hInstance;


wc.hIcon = LoadIcon(NULL, "boope.ico");//


wc.hCursor = LoadCursor(NULL, IDC_ARROW);


wc.hbrBackground = GetStockObject(WHITE_BRUSH);


wc.lpszMenuName = NULL;


wc.lpszClassName = "APP";


return(RegisterClass(&wc));

}

BOOL IniciaInstancia(HANDLE hInstance, int nCmdShow)

{


hWnd = CreateWindow(



"APP",



"Aplicación sencilla",



WS_OVERLAPPEDWINDOW,



10,



10,



300,



80,



NULL,



NULL,



hInstance,



NULL);


if (!hWnd)



return (FALSE);


ShowWindow(hWnd, nCmdShow);


UpdateWindow(hWnd);


return (TRUE);

}

int PASCAL WinMain(HANDLE hInstance, HANDLE hPrevInstance, LPSTR lpCmdLine, int nCmdShow)

{


MSG Mensaje;


if (!hPrevInstance){



if (!IniciaAplicacion(hInstance))




return (FALSE);


}


if (!IniciaInstancia(hInstance,nCmdShow))



return (FALSE);


while (GetMessage(&Mensaje,NULL,NULL,NULL)){



TranslateMessage(&Mensaje);



DispatchMessage(&Mensaje);


}


return (Mensaje.wParam);

}

Luego en el menú seleccionan “Build” y le dan clic a la opción ”Build archivo.exe” o sencillamente le dan clic a un signo de exclamación rojo que aparece en la barra de herramienta. Si todo funciona a la perfección verán un simple formulario parecido al de VB pero con un fondo blanco.

Señores, les presento con este código la forma mas “simple” o la forma mas primitiva de crear un formulario, lenguajes como VC++ (con las librerías MFC), Visual Foxpro, VJ++, y VB, hacen esto detrás de “vestidores”, este es el código para crear un formulario simple en Windows. ¿Por que les presento este código?, por que con este código les explicare cierto protocolo que sigue Windows para la iniciación de un programa. Olvídense que te escrito en lenguaje C nos centraremos en su estructura.

Empezamos con:

#include <windows.h>

Esta instrucción abre el archivo el cual contiene todas las referencias a las funciones API o bueno no todas pero las mayoría, aquí se encuentran las funciones API básicas y la mayoría de los mensajes.

Después toca explicar esto:

int PASCAL WinMain(HANDLE hInstance, HANDLE hPrevInstance, LPSTR lpCmdLine, int nCmdShow)

{


MSG Mensaje;


if (!IniciaAplicacion(hInstance))




return (FALSE);


if (!IniciaInstancia(hInstance,nCmdShow))



return (FALSE);


while (GetMessage(&Mensaje,NULL,NULL,NULL)){



TranslateMessage(&Mensaje);



DispatchMessage(&Mensaje);


}


return (Mensaje.wParam);

}

Para los conocedores de C, C++ o Java, saben que existe una función de arranque que se llama main, la cual es la primera función que arranca cuando se inicia el programa, en Windows su equivalente es WinMain. Esta función le pasa a la aplicación 4 cosas, la primera “hInstance” es un número que Windows le da a tu aplicación, ese número lo usamos mas adelante, la segunda “hPrevInstance”, nos dice si hay otra instancia abierta de nuestra aplicación, esto es por si ejecutamos la misma aplicación 2 veces, o 3, etc. “LpCmdLine” son los parámetros que le pasamos a nuestra aplicación cuando arranca, se recuerdan de DOS cuando hacíamos esto:

Programa .exe /c /d /e, bueno este parámetro contienes las letras c,d,e.

Y por último tenemos “nCmdShow” que es el como se va abrir nuestra aplicación, si maximizada, minimizada o normal. Parecido al parámetro “Windows State” de VB.

Lo primero que hacemos es iniciar la aplicación.

if (!IniciaAplicacion(hInstance))




return (FALSE);

Función:

BOOL IniciaAplicacion(HANDLE hInstance)

{


WNDCLASS wc;


wc.style = CS_HREDRAW | CS_VREDRAW ;


wc.lpfnWndProc = GestorMensajes;


wc.cbClsExtra = 0;


wc.cbWndExtra = 0;


wc.hInstance = hInstance;


wc.hIcon = LoadIcon(NULL, IDI_APPLICATION);//


wc.hCursor = LoadCursor(NULL, IDC_ARROW);


wc.hbrBackground = GetStockObject(WHITE_BRUSH);


wc.lpszMenuName = NULL;


wc.lpszClassName = "APP";


return(RegisterClass(&wc));

}

aquí creamos un objeto WNDCLASS este objeto se utiliza para configurar la apariencia de nuestra ventana futura que todavía no ha sido creada.


wc.style = CS_HREDRAW | CS_VREDRAW ;  aquí lo que estamos configurando es que cuando la ventana se redimensione tanto vertical como horizontal, se repinte de nuevo.


wc.lpfnWndProc = GestorMensajes; aquí estamos configurando nuestro “oido”, es decir, colocamos el nombre de la función encargada de recibir todos los mensajes que se le envíen a nuestra ventana.


wc.cbClsExtra = 0;


wc.cbWndExtra = 0; Parámetros no importantes.

wc.hInstance = hInstance; Aquí colocamos el  número que nos dio Windows. Cuando se inicia la aplicación. OJO! Este no es todavía el HWND.


wc.hIcon = LoadIcon(NULL, IDI_APPLICATION); Aquí colocamos el icono que queremos,  por defecto elegimos uno del sistema.

wc.hCursor = LoadCursor(NULL, IDC_ARROW); Aquí colocamos el tipo de puntero que queremos para nuestra aplicación.

wc.hbrBackground = GetStockObject(WHITE_BRUSH); Asignamos el color de fondo.


wc.lpszMenuName = NULL; Indicamos si tiene menú o no, en caso de haber menú se le coloca el nombre del recurso. 
wc.lpszClassName = "APP";  Indicamos el nombre de la clase, esto de la clase lo veremos mejor cuando veamos una API relacionada con ella.


return(RegisterClass(&wc)); Ya con los parámetros configurados le pedimos a Windows con esta función API RegisterClass, que registre nuestra configuración y si es valida retorna  TRUE y si es FALSE pues implica que nos equivocamos en los parámetros o cometimos algún error.

Después sigue la siguiente línea.


if (!IniciaInstancia(hInstance,nCmdShow))



return (FALSE);

Con nuestra instancia ya registrada, procedemos a crear físicamente nuestra ventana.

BOOL IniciaInstancia(HANDLE hInstance, int nCmdShow)

{


hWnd = CreateWindow(



"APP",



"Aplicación sencilla",



WS_OVERLAPPEDWINDOW,



10,



10,



300,



80,



NULL,



NULL,



hInstance,



NULL);


if (!hWnd)



return (FALSE);


ShowWindow(hWnd, nCmdShow);


UpdateWindow(hWnd);


return (TRUE);

}

Aquí es donde realmente creamos ya nuestra ventana físicamente, para ello usamos la API CreateWindow. Creo que el nombre de la función API representa de manera obvia lo que hace.


hWnd = CreateWindow(



"APP", Nombre de la clase que usamos en el objeto WC de arriba


"Aplicación sencilla", Nombre del Caption o de la ventana


WS_OVERLAPPEDWINDOW, Estilo de la ventana


10, Posición X de la ventana



10, Posición Y de la ventana


300, Ancho


80, Alto


NULL,



NULL,



hInstance, Instancia a la cual corresponde.



NULL);

Y esta función es la que nos regresa nuestro HWND correspondiente. Luego si el HWND es valido. Procedemos a Mostrar la ventana


ShowWindow(hWnd, nCmdShow);

Noten que ya a partir de este momento usamos nuestro HWND, y le indicamos como se mostrara nuestra ventana.

Y luego le decimos que se pinte con:


UpdateWindow(hWnd);

Y listo ya tenemos nuestra primera ventana hecha a “mano”.

Quienes hayan visto sistema operativo en la universidad aquí esta el bucle de mensajes , el cual se encarga de escuchar los mensajes que recibe la ventana, toda ventana internamente tiene esto o algo parecido, lo que sucede es que en los otros lenguajes no se ve.


while (GetMessage(&Mensaje,NULL,NULL,NULL)){



TranslateMessage(&Mensaje);



DispatchMessage(&Mensaje);


}

LRESULT CALLBACK GestorMensajes(HANDLE hWnd, UINT mensaje, WPARAM wParam, LPARAM lParam)

{


switch (mensaje) {


case WM_DESTROY:



PostQuitMessage(0);



break;


default:



return DefWindowProc(hWnd,mensaje,wParam,lParam);


}

}

Esta ultima función es la encargada de escuchar, ella es nuestro oido, ella recibirá TODOS LOS MENSAJES que se generen para esta ventana y luego quedara en nosotros la misión de codificar cada mensaje si fuera necesario. Se recuerdan de la estructura:

Componente que recibe el mensaje, Tipo de mensaje, mensaje1, mensaje_opcional
Esta estructura se convierte en esto:

HANDLE hWnd, UINT mensaje, WPARAM wParam, LPARAM lParam

Algunos de ustedes, pueden que no hayan entendido, y yo les digo ¡No importa!, ya que esto no lo vamos a usar, solo se los mostré como cultura general para que vean lo que hay detrás de esos formularios ya creados por diferentes lenguajes, y vean que todo el formulario se crea con funciones API de Windows. Lo que quiero que vean es que TODO gira alrededor de las API de Windows, no puedes hacer una aplicación para Windows sin usarlas. En este momento ustedes dicen: ¡ESO ES MENTIRA!, yo hice una aplicación en VB y no use API, OK! Tienes razón pero internamente ese botón que usaste, ese listbox y el propio formulario internamente están usando las API de Windows para su funcionamiento. Lo que sucede es que programar a este nivel es demasiado complicado!,  inclusive si tu me preguntaras que porcentaje del 1 al 100% consideras tus conocimientos en API,  yo te diría en un 5 al 7%, ya que esto es extremadamente grande!!!. 

Lo que hacen los programadores y las empresas es crear controles y librerías que esconden la parte fea del asunto, poniéndolo todo bonito para que sea fácil para el programador manipularlas y así diseñar los software de manera rápida y efectiva.

¿CÓMO USAR API’S WIN32 EN VB?.

Para usar las API Win32 en VB hay que hacer ciertas cositas, si vas a declarar las API’S en un modulo tiene que ser de esta manera:

Declare Function ShowWindow Lib "user32" (ByVal hwnd As Long, ByVal nCmdShow As Long) As Long

Primero se coloca Declare Function, luego el nombre de la función API, después lib y entre comillas se coloca el archivo .dll que contiene la función, y luego dependiendo del tipo de función se le coloca sus respectivos parámetros. Cada función API se encuentra en la ayuda de Microsoft en www.msdn.microsoft.com o en la ayuda MSDN del visual studio.

En caso de usar la API en un formulario se le coloca la palabra PRIVATE antes de DECLARE.

NUESTRA PRIMERA FUNCIÓN API SENDMESSAGE

En esta sección vamos a ver nuestra primera API, que es el SendMessage, como todos sabemos la acción que usa un OBJETO para comunicarse con otro OBJETO es el MENSAJE, ese mensaje lleva una información que se procesará en el OBJETO receptor.

El API se compone de esta manera:

SendMessage (hwnd,wMsg,wParam, lParam)

Estudiemos su estructura:

hwnd = es ese número que le asigna WINDOW a ese objeto o ventana receptora.

wMsg = es el Mensaje.

wParam = es la información  que se le envía al hwnd.

lParam = es otra información que también se le envía al hwnd.

En Windows los tipos de mensajes vienen definidos por CONSTANTES. Existen miles de constantes y tipos de mensaje que se pueden enviar a un objeto o hwnd. Ahora hay ciertos mensaje que están definidos para cierto tipo de ventana. Por ejemplo hay mensajes solo para TEXTBOX, otras solo para LISTBOX, etc.

Veamos nuestro primer programa en Visual Basic. Para este ejemplo es necesario tener un botón en un formulario. A continuación copien el siguiente código:

Private Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Const WM_SETTEXT = &HC

Private Sub Command1_Click()

Dim Resultado As Long

Resultado = SendMessage(Command1.hwnd, WM_SETTEXT, 0, ByVal "GRACIAS")

End Sub

Private Sub Form_Load()

    Command1.Caption = "PRESIONAME"

End Sub

ANALISIS

Private Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Van a notar en los parámetros que se le pasan a la función, que tienen la palabra ByVal, esto significa que los parámetros se pasan por valor y no por referencia. He de suponer que ustedes a este nivel tienen claro esos términos.

El “lParam as Any” el Any es que acepta cualquier objeto en este caso una cadena “GRACIAS”.

Con este código verán que cuando le dan click al botón, cambia y dice “GRACIAS”, ahora que sucedió:

Resultado = SendMessage(Command1.hwnd, WM_SETTEXT, 0, ByVal "GRACIAS")

RESULTADO = Todos los mensaje o en su mayoría retornan VALORES, siempre todas las funciones API regresan un número de tipo LONG. En este caso no interesa el valor a retornar pero más adelante veremos que sí.


HWND = COMMAND1.hwnd

Este es nuestro objeto receptor, es decir el objeto a quién le estamos enviando el mensaje.


MENSAJE = WM_SETTEXT

Mensaje que declaramos como constante en el área de “DECLARACIONES”, normalmente estos son valores hexadecimales.


Wparam = 0 

En esta caso aquí no va el mensaje.


LParam = “ ByVal “GRACIAS”


El ByVal es muy importante ya que obliga a que se pase el valor por copia y no por referencia. Y este es el mensaje que le pasamos.

MAS SOBRE MENSAJES.

Si ustedes cambian el Command1.hwnd y colocan Me.hwnd y lo corren cuando aprieten el botón van a ver el GRACIAS, no en el botón sino en el CAPTION del formulario, esto es por que  el mensaje WM_SETTEXT sirve para cualquier hwnd. Ahora como saber que mensaje sirve para tal objeto.

Bueno aquí te muestro una lista de algunos de ellos:


BM_



Button mensaje


CB_



Combo Box Mensaje


CDM_



Common Dialog Mensaje


DRV_



Driver Mensaje


EM_



Edit control o TextBox Mensaje


LB_



List Box Mensaje


WM_



General Window Mensaje


Etc.

Cuando ustedes vean un mensaje o constante que empiece por BM_ significa que es un mensaje para un botón si lo usan en un Combo Box no funcionará. Es por esa razón que WM_SETTTEXT funcionó tanto para el botón como para el formulario.

Tanto para los mensajes como para la funciones tienen que tomar en cuenta un detalle, vean que el mensaje es WM_SETTEXT el SET se usa cuando asignamos y el GET para tomar un valor, es decir, cuando vean una función:


SETXNOMBRE = significa que vamos asignar algo.


GETXNOMBRE = significa que vamos a obtener algo.

En este caso el WM_ = mensaje que identifica que es un hwnd o ventana, SET = asignamos, TEXT = texto.

En este caso el TEXTO de la VENTANA botón es lo que llamamos en VB la propiedad CAPTION, al igual con el formulario.

Ahora vamos a probar algo interesante pero con GET

Declara este mensaje 

Const WM_GETTEXT = &HD

Luego en el evento click del botón cambiamos el código que esta por el siguiente:

Private Sub Command1_Click()

Dim Resultado As Long

Resultado = SendMessage(Command1.hwnd, WM_GETTEXT, 0, 0)

End Sub

Arranca y presiona el botón verás que no cambia nada, ni se ve nada, tu dirás bueno el Texto del hwnd debe de estar en la variable “Resultado”, bueno vamos a probar pon debajo de:

   Resultado = SendMessage(Command1.hwnd, WM_GETTEXT, 0, 0)

   Msgbox Resultado

Cuando lo corras verás que te muestra 0, las funciones API de WINDOW retornan nada mas resultados de tipo LONG, no retornan ni STRING, ni FLOAT, etc. Ustedes dirán ¿bueno y como hacemos?. Cuando un API retorna una información ya sea de texto o una estructura (ya veremos eso mas adelante) se usan de esta manera:

Cambia el código en el evento Click del botón por este:

Private Sub Command1_Click()

Dim Resultado As Long

Dim Texto As String

Texto = Space$(20)

Resultado = SendMessage(Command1.hwnd, WM_GETTEXT, 15, ByVal Texto)

MsgBox Texto

End Sub


Dim Texto as String

Verás que declaramos una variable de tipo String

Texto = Space$(20)

Le asignamos 20 espacios en blancos a la variable, cuando quieres llenar una variable de tipo STRING sea cual sea la función API, tienes que asignarle el espacio en memoria, eso lo hacemos asignándole 20 espacios en blanco, pero el  número puede variar.

SendMessage(Command1.hwnd, WM_GETTEXT, 15, ByVal Texto)

En esta función lo que cambia es el mensaje y los parámetros, el mensaje cambia a GETTTEXT es decir, “tomar el texto”, wParam = 15 esto significa que le dices a WINDOW que te regrese 15 caracteres nada más, si la cadena a regresar es mayor a 15 te le regresa incompleta. lParam = ByVal Texto; “Texto” es la variable donde se almacenará el Texto que Windows regrese. Si tu declaras por ejemplo:

Texto = Space$(2)

Cuando Windows intente llenar la variable no lo va llenar con el texto completo y te puede aparecer caracteres extraños. Esto es por que cuando quiere grabar el Texto del hwnd en la memoria se encuentra con una variable que fue inicializada con dos caracteres. 

Normalmente se declaran las variables con 255, ejemplo, Texto = Space$(255), luego si quieres eliminar los espacios en blancos usamos la función TRIM de VB para ello.

El SendMessage es una de las API mas usadas en la programación bajo WINDOW, bueno es obvio que nosotros nunca vamos hacer lo que hicimos arriba para asignarles un Texto a una hwnd (Verán que uso la palabra hwnd y no ventana esto es por que no quiero que se confundan de que ventana es el Formulario, necesito que se acostumbre a que un hwnd sea cual fuera, es una ventana). Esto es por que los controles de VB resumen ese paso en sus propiedades, es decir, cuando cambias el texto a un hwnd (botón) lo que haces es usar su propiedad CAPTION. Obviamente ya ustedes se imaginarán lo que hace CAPTION en su interior.

La finalidad de explicar todo lo que dije arriba, fue para que vean como se maneja un API, ya que las API’S en su manejo son similares dependiendo de su tipo. A partir de ahora vamos a empezar a ver API’S un poco más complejas, ya que entenderlas implica entender a WINDOW y eso es un poco complicado. VB es un lenguaje muy sencillo y tal vez necesiten de las API’S para problemas extremos, pero les aseguro que si quieren aprender lenguajes como VC++ necesitarán del mas mínimo conocimiento de API’S para entenderlo.

SENDMESSAGE Y POSTMESSAGE

PostMessage es exactamente igual a SendMessage, la diferencia radica en que cuando mandan mensaje con SendMessage el código que esta inferior a él no se ejecuta hasta que termine, por ejemplo:

Private Sub Command1_Click()

    Dim Resultado As Long

    Dim Texto As String

    Texto = Space$(20)

    Resultado = SendMessage(Command1.hwnd, WM_GETTEXT, 15, ByVal Texto)

    MsgBox Texto ‘ Esta línea no se ejecuta hasta que el SendMessage se ejecute por completo

End Sub

Si usamos PostMessage:

Private Sub Command1_Click()

    Dim Resultado As Long

    Dim Texto As String

    Texto = Space$(20)

    Resultado = PostMessage(Command1.hwnd, WM_GETTEXT, 15, ByVal Texto)

    MsgBox Texto ‘ Esta línea se ejecuta sin necesidad de que PostMessage haya finalizado.    

End Sub

El uso de PostMessage no es muy usual, inclusive todavía no he visto un ejemplo práctico con esta función. Pero ya saben que existe.

NOTA: Ustedes se preguntaran, ¿COMO SABER CUANDO USAR WPARAM Y LPARAM? o ¿COMO SE QUE VALOR LES PASO?, para ello no existe ninguna regla, cuando ustedes quieran usar un mensaje con SendMessage ustedes tienen que buscar en la ayuda de MSDN que parámetros se le envían a ese mensaje.

WINDOWS MESSAGE

En esta sección vamos a ver algunos mensajes útiles para las ventanas o hwnd usando el SendMessage, es importante mencionar, que no voy a explicar todos los mensajes para cada control, sino los que yo considero más importantes, recuerden que esta manual no esta orientado, para ser un diccionario de API sino explicar como funcionan mediante ejemplos.

EDIT CONTROL O TEXTBOX

· Mensaje : EM_GETLINE

Valor : Const EM_GETLINE = &HC4

Mensaje que funciona en un Textbox Multiline, este permite tomar el Texto que se encuentra en una línea específica.

Dim Respuesta As Long

Dim Texto As String

Texto = Space$(10)

Respuesta = SendMessage(Text1.hwnd, EM_GETLINE, 1, ByVal Texto)

MsgBox Texto

Lo único que hay que explicar es el valor que toma wParam que es 1, este es el número de línea de la cual queremos ver el texto. Si ponemos 0 sería la primera línea, 1 la segunda línea, así sucesivamente.

· Mensaje: EM_GETLINECOUNT

Valor: Const EM_GETLINECOUNT = &HBA

Mensaje que sirve para retornar la cantidad de líneas que posee un Textbox configurado de manera multiline.

Dim Respuesta As Long

    Respuesta = SendMessage(Text1.hwnd, EM_GETLINECOUNT, 0, 0)

    MsgBox Respuesta

Si vemos aquí, si nos interesa el valor de retorno del Mensaje, verán que no se le pasa ningún parámetro o información, con el tipo de Mensaje basta para que él sepa a que nos referimos. El número de líneas se almacena en la variable Respuesta.

· Mensaje: EM_GETMODIFY

Valor: Const EM_GETMODIFY = &HB8

Esta mensaje sirve para saber si un TEXTBOX, multiline o simple, se ha modificado. 

Dim Respuesta As Long

 Respuesta = SendMessage(Text1.hwnd, EM_GETMODIFY, 0, 0)

       MsgBox Respuesta

Los valores que retornan es 0 si no se ha modificado y 1 si fue modificado.

· Mensaje : EM_LINELENGTH

Valor: Const EM_LINELENGTH = &HC1

Este mensaje sirve para tomar la longitud de una línea de un TEXTBOX sea multiline o no.

    Dim Respuesta As Long

    Respuesta = SendMessage(Text1.hwnd, EM_LINELENGTH, 1, 0)

    MsgBox Respuesta

Aquí el mensaje retorna la longitud sería como el LEN de VB. wParam es el número de la línea de la cual queremos tomar su longitud. Esto si el Textbox es Multiline, en caso de no serlo no importa el número que pongas en wParam, siempre te retornará la longitud de la única línea que tiene.

· Mensaje : EM_SETSEL 
Valor: Const EM_SETSEL = &HB1

Esta mensaje sirve para seleccionar una porción de la cadena que esta contenida en el TEXTBOX.

SendMessage Text1.hwnd, EM_SETSEL, 5, 1

MsgBox Text1.SelText

Aquí wParam recibe la posición donde se inicia la selección, el texto seleccionado se encontrara en la Propiedad .SelText del Textbox. 

Noten que esta función la invocamos diferente. Cuando el valor que retorna una función no interesa, podemos usar la función como que si llamáramos a un método, Por ejemplo:

Respuesta = SendMessage(Text1.hwnd, EM_LINELENGTH, 1, 0)

En este caso nos interesa el resultado de la función por que la variable “Respuesta” toma el valor de la longitud de la cadena.

Respuesta = SendMessage(Text1.hwnd, EM_SETSEL, 5, 1)

Aquí “respuesta” no nos sirve para nada, por consiguiente, cuando el valor de la función sea cual sea el API no nos interese, podemos usarla de esta manera:

SendMessage Text1.hwnd, EM_SETSEL, 5, 1

Siendo completamente válido esta instrucción. Pero tienen que tomar en  cuenta que algunas veces lo que retorna una función no lo usamos con un propósito, pero algunas veces el resultado que devuelve una función puede estar diciendo si la operación se ejecuto o no se ejecuto con éxito. Les repito eso dependerá del mensaje que ustedes utilicen.

COMBO BOX CONTROL

· Mensaje: CB_DIR

Valor : Const CB_DIR = &H145

Este mensaje, el cual lo veo muy importante permite llenar un COMBO BOX con información del sistema. En este caso a wParam le pasaremos unos parámetros ya predefinidos por WINDOW.

Parámetros que se pasan por Wparam:

Const DDL_ARCHIVE = &H20

Const DDL_DIRECTORY = &H10

Const DDL_DRIVES = &H4000

Const DDL_EXCLUSIVE = &H8000

Const DDL_HIDDEN = &H2

Const DDL_READONLY = &H1

Const DDL_READWRITE = &H0

Const DDL_SYSTEM = &H4

Ejemplo:

Dim path As String    

path = App.path

SendMessage Combo1.hwnd, CB_DIR, DDL_DRIVES, ByVal path

Se darán cuenta que al ejecutarlo, todas las letras y unidades de la computadora se almacenan en el combo, verán que en wParam  toma el parámetro que declaramos como constante DDL_DRIVES, los parámetros también pueden ser constantes ya establecidas o que poseen un valor. En lParam verán que es la ruta de nuestro programa, que es lo que exige el mensaje como segundo parámetro. Para saber que se les pasa a los otros parámetros que van en wParam, les recomiendo que busquen la ayuda en Línea de MSDN, ya que la ayuda normal, no muestra que parámetros son.

· Mensaje : CB_FINDSTRING

Valor: Const CB_FINDSTRING = &H14C

Este mensaje muy importante en el momento de hacer búsqueda en un COMBO ya que permite localizar cierto registro en el COMBO. En este mensaje wParam es el índice en donde empezamos a buscar y lParam es la cadena que buscamos.

    Dim Texto As String

    Dim  Respuesta As Long

    Texto = "Prueba"

    Respuesta = SendMessage(Combo1.hwnd, CB_FINDSTRING, -1, ByVal Texto)

    MsgBox Respuesta

Podremos ver que wParam le coloque el valor de –1, eso es para que busque en toda la lista del combo, si pongo por ejemplo 2 en wParam, empezará a buscar desde el  registro 3, recuerden que el combo empieza en el índice 0.

Si la función encuentra la cadena regresa el índice en donde se encuentra en el combo, en caso de que no exista regresa –1. En esta función la búsqueda no es exacta, es decir:

“PRUEBA” Imagínense que tienen esta cadena en el Combo, si ustedes asignan la variable “Texto” de esta manera:


Texto = “Pru”

Cuando realice la búsqueda para el tipo de mensaje “PRUEBA” es igual a “PRU”, esto es por que  la cadena en el combo empieza por “PRU”. Para hacer una búsqueda exacta vean el mensaje a continuación.

· Mensaje : CB_FINDSTRINGEXACT

Valor : Const CB_FINDSTRINGEXACT = &H158

Esta mensaje es igual al anterior con la diferencia que busca la cadena exacta en la lista del combo.

    Dim Texto As String

    Dim Respuesta As Long

    Texto = "Prueba"

    Respuesta = SendMessage(Combo1.hwnd, CB_FINDSTRINGEXACT, -1, ByVal Texto)

    MsgBox Respuesta

Si seguimos el ejemplo de “PRU”, para este mensaje “PRU” no es igual a “PRUEBA”, ya que esta función busca la cadena que sea EXACTAMENTE igual.

Estas dos funciones son muy importantes en el momento de realizar búsquedas en COMBO que contengan mucha información, es común de los programadores en VB hacer:


For i = 0 to Combo1.listcount –1



Código para buscar


Next

Usando el SendMessage les aseguro que van a tener resultados increíblemente rápidos.

· Mensaje : CB_SHOWDROPDOWN

Valor : Const CB_SHOWDROPDOWN = &H14F

Esta función se utiliza para mostrar la lista desplegable del combo box.


SendMessage Combo1.hwnd, CB_SHOWDROPDOWN, True, 0

En este caso el valor de wParam toma el valor TRUE para mostrar la lista y FALSE para esconderla.

Capítulo  III.  API’S WIN32 GRAFICAS (GDI)

Windows a nivel grafico usa la GDI (Graphics Device Interface o Interfaz grafica de dispositivo), la GDI no es mas que un conjunto de librerías las cuales se usan para la graficación en Windows, es curiosos pero si pensamos un poquito todo gira en la GDI, claro es como obvio, ya que todo Windows es grafico, estas funciones o librerías se encuentran en su mayoría en el archivo gdi32.dll, con estas funciones podemos dibujar formularios, botones, textos, rayas, formas, etc. Pero vamos a pensar un poco mas abstracto, y no ver la GDI como las funciones usadas para obtener un resultado para presentar en el monitor, una hoja de papel impresa en color también es grafica por lo tanto la GDI juega un papel importante.

Ahora como juega nuestro HDC con todo esto, imaginemos a un pintor, básicamente que tiene: un cabestrillo, un papel, un pincel, y colores, bueno nuestro cabestrillo es el HWND, es nuestro soporte donde nosotros ponemos nuestro papel que sería el HDC, el HDC es el papel donde nosotros dibujamos y ese papel puede ser enviado a una pantalla o impresora, o cualquier dispositivo de salida grafico. Claro el reto esta en ¿Cómo hacemos para pintar en nuestra hoja?. Voy a mostrar a continuación herramientas de dibujo útiles para el desarrollo de nuestros proyectos, claro esta, que Visual Basic esta parte la esconde en su interior ya que algunas cosas suelen ser un poco tediosas, pero es claro entender que todo lo que hablaremos aquí esta relacionado con nuestra hoja (HDC) donde pondremos nuestra imaginación a correr.

BROCHAS

Volviendo a pintor, imaginemos que necesita pintar un mural en una calle, para el fondo de su pintura es claro pensar que agarrara una brocha, pintara el fondo de un color y luego continuara el resto,  ¿es esto algo parecido a la computación?,  claro que si, una brocha no es mas que pintar el fondo de un HDC o parte del fondo, usando una forma geométrica para ello.

Pero vamos a lo importante, abran un proyecto en VB (Visual Basic) luego coloquen un listbox, el listbox aléjenlo del borde izquierdo superior (0,0), es decir, colóquenlo  en left = 3000, y top = 0, la imagen bloque.bmp que viene con este manual colóquenla en donde esta el proyecto y luego coloquen este código:

Private Declare Function CreateHatchBrush Lib "gdi32" (ByVal nIndex As Long, ByVal crColor As Long) As Long

Private Declare Function CreatePatternBrush Lib "gdi32" (ByVal hBitmap As Long) As Long

Private Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Private Declare Function GetSysColorBrush Lib "user32" (ByVal nIndex As Long) As Long

Private Declare Function GetStockObject Lib "gdi32" (ByVal nIndex As Long) As Long

Private Declare Function GetSysColor Lib "user32" (ByVal nIndex As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Const HS_BDIAGONAL = 3

Private Const HS_CROSS = 4

Private Const HS_DIAGCROSS = 5

Private Const HS_FDIAGONAL = 2

Private Const HS_HORIZONTAL = 0

Private Const HS_VERTICAL = 1

Private Const BLACK_BRUSH = 4

Private Const DKGRAY_BRUSH = 3

Private Const GRAY_BRUSH = 2

Private Const HOLLOW_BRUSH = 5

Private Const LTGRAY_BRUSH = 1

Private Const WHITE_BRUSH = 0

Private Const COLOR_3DDKSHADOW = 21

Private Const COLOR_3DLIGHT = 22

Private Const COLOR_BTNFACE = 15

Private Const COLOR_BTNHIGHLIGHT = 20

Private Const COLOR_BTNSHADOW = 16

Private Const COLOR_ACTIVEBORDER = 10

Private Const COLOR_ACTIVECAPTION = 2

Private Const COLOR_BACKGROUND = 1

Private Const COLOR_BTNTEXT = 18

Private Const COLOR_CAPTIONTEXT = 9

Private Const COLOR_GRADIENTACTIVECAPTION = 27

Private Const COLOR_GRADIENTINACTIVECAPTION = 28

Private Const COLOR_GRAYTEXT = 17

Private Const COLOR_HIGHLIGHT = 13

Private Const COLOR_HIGHLIGHTTEXT = 14

Private Const COLOR_HOTLIGHT = 26

Private Const COLOR_INACTIVEBORDER = 11

Private Const COLOR_INACTIVECAPTION = 3

Private Const COLOR_INACTIVECAPTIONTEXT = 19

Private Const COLOR_INFOBK = 24

Private Const COLOR_INFOTEXT = 23

Private Const COLOR_MENU = 4

Private Const COLOR_SCROLLBAR = 0

Private Const COLOR_WINDOW = 5

Private Const COLOR_WINDOWFRAME = 6

Dim brocha As Long

Private Sub Form_Load()


Me.AutoRedraw = True

    List1.AddItem "HATCH BRUSH - HS_DIAGONAL"

    List1.AddItem "HATCH BRUSH - HS_CROSS"

    List1.AddItem "HATCH BRUSH - HS_DIAGROSS"

    List1.AddItem "HATCH BRUSH - HS_FDIAGONAL"

    List1.AddItem "HATCH BRUSH - HS_HORIZONTAL"

    List1.AddItem "HATCH BRUSH - HS_VERTICAL"

    List1.AddItem "SOLID BRUSH - BLACK_BRUSH"

    List1.AddItem "SOLID BRUSH - DKGRAY_BRUSH"

    List1.AddItem "SOLID BRUSH - GRAY_BRUSH"

    List1.AddItem "SOLID BRUSH - HOLLOW_BRUSH"

    List1.AddItem "SOLID BRUSH - LTGRAY_BRUSH"

    List1.AddItem "SOLID BRUSH - WHITE_BRUSH"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_3DDKSHADOW"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_3DLIGHT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_BTNFACE"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_BTNHIGHLIGHT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_BTNSHADOW"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_ACTIVEBORDER"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_ACTIVECAPTION"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_BTNTEXT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_CAPTIONTEXT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_GRADIENTACTIVECAPTION"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_GRADIENTINACTIVECAPTION"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_GRAYTEXT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_HIGHLIGHT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_HIGHLIGHTTEXT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_HOTLIGHT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_INACTIVEBORDER"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_INACTIVECAPTION"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_INACTIVECAPTIONTEXT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_INFOBK"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_INFOTEXT"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_MENU"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_SCROLLBAR"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_WINDOW"

    List1.AddItem "SOLID COLOR SYSTEM - COLOR_WINDOWFRAME"

    List1.AddItem "BRUSH PATTERN - IMAGEN"

End Sub

Private Sub Form_Unload(Cancel As Integer)

    DeleteObject brocha

End Sub

Private Sub List1_Click()

    Select Case List1.ListIndex

    Case 0

        Me.Cls

        brocha = CreateHatchBrush(HS_DIAGONAL, RGB(0, 0, 0))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 1

        Me.Cls

        brocha = CreateHatchBrush(HS_CROSS, RGB(0, 0, 0))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 2

        Me.Cls

        brocha = CreateHatchBrush(HS_DIAGROSS, RGB(0, 0, 0))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 3

        Me.Cls

        brocha = CreateHatchBrush(HS_FDIAGONAL, RGB(0, 0, 0))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 4

        Me.Cls

        brocha = CreateHatchBrush(HS_HORIZONTAL, RGB(0, 0, 0))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 5

        Me.Cls

        brocha = CreateHatchBrush(HS_VERTICAL, RGB(0, 0, 0))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 6

        Me.Cls

        SelectObject Me.hdc, GetStockObject(BLACK_BRUSH)

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 7

        Me.Cls

        SelectObject Me.hdc, GetStockObject(DKGRAY_BRUSH)

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 8

        Me.Cls

        SelectObject Me.hdc, GetStockObject(GRAY_BRUSH)

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 9

        Me.Cls

        SelectObject Me.hdc, GetStockObject(HOLLOW_BRUSH)

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 10

        Me.Cls

        SelectObject Me.hdc, GetStockObject(LTGRAY_BRUSH)

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 11

        Me.Cls

        SelectObject Me.hdc, GetStockObject(WHITE_BRUSH)

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 12

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_3DDKSHADOW))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 13

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_3DLIGHT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 14

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_BTNFACE))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 15

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_BTNHIGHLIGHT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 16

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_BTNSHADOW))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 17

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_ACTIVEBORDER))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 18

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_ACTIVECAPTION))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 19

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_BACKGROUND))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 20

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_BTNTEXT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 21

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_CAPTIONTEXT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 22

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_GRADIENTACTIVECAPTION))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 23

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_GRADIENTINACTIVECAPTION))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 24

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_GRAYTEXT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 25

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_HIGHLIGHT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 26

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_HIGHLIGHTTEXT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 27

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_HOTLIGHT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 28

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_INACTIVEBORDER))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 29

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_INACTIVECAPTION))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 30

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_INACTIVECAPTIONTEXT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 31

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_INFOBK))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 32

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_INFOTEXT))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 33

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_MENU))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 34

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_SCROLLBAR))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 35

        Me.Cls

        brocha = CreateSolidBrush(GetSysColor(COLOR_WINDOW))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    Case 36

        Me.Cls

        brocha = CreatePatternBrush(LoadPicture(App.Path & "\bloque.bmp"))

        SelectObject Me.hdc, brocha

        Rectangle Me.hdc, 0, 0, 100, 100

        Me.Refresh

    End Select

End Sub

Cuando arranquen la aplicación verán que el listbox se llena con valores y al seleccionar cada opción aparece un recuadro en la esquina superior izquierda que ve cambiando de color y de fondo a medida que nos movemos por el listado, en este momento ustedes dirán ¿Cómo se hace todo esto?.

Bueno primero que nada vamos explicar que existen 3 tipos de brochas, las sólidas, las tramadas y la texturizada.

SOLIDAS

Esta se encarga de crear una brocha de un color sólido. Como se puede notar viendo su API.

CreateSolidBrush

Api: Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Pueden ver que como único parámetro toma un color, pueden usar la función RGB para ello,  mas adelante les explicare lo que hice.

TRAMADAS

No es mas que crear una brocha con tramas ya definidas, estas la vemos en las primeras 6 opciones. 

CreateHatchBrush

Api: Declare Function CreateHatchBrush Lib "gdi32" (ByVal nIndex As Long, ByVal crColor As Long) As Long

Esta API en este caso toma un color, y un “nIndex”, este nIndex pueden ser 6 valores, que son:

Private Const HS_BDIAGONAL = 3

Private Const HS_CROSS = 4

Private Const HS_DIAGCROSS = 5

Private Const HS_FDIAGONAL = 2

Private Const HS_HORIZONTAL = 0

Private Const HS_VERTICAL = 1

Y es fácil ver en el programa la forma del tramo que forma cada constante, y el color dependerá, de como configuren la variable “crColor” de la función.

Texturizadas

Con esta creamos una brocha que en ves de tener color tiene una textura (imágenes), lamentablemente en Window95/98/ME solamente se pueden usar texturas 8x8 píxeles, no pueden usar una mas grande ya que en caso de colocar una mas grande Windows les tomara los primeros 8x8 píxeles, en caso de trabajar con Window2000/NT pueden usar imágenes de tamaño indefinido. Y supongo que XP debe de tener también esa posibilidad.

Ustedes dirán: hago el programa en Windows 2000, uso una imagen grande la cual funciona y crees que al compilar el .exe en Windows 2000 no vas tener problema con Windows 98 o 95, déjame decirte que eso no funciona así!, el hecho que lo compiles en Windows 2000 no implica que te funcionara en Windows 98 o 95, ya que el gdi32.dll en Windows 2000 es distinto que el de Windows 98 por mas que se llamen igual. Y NI SE LES OCURRA suplantar el gdi32.dll del Windows 98 con el de Windows 2000, creyendo que así solucionan el problema, por que lo que van a ganar es que Windows se ponga un poco loco y eso no lo queremos. Ahora veamos el API involucrada.

CreatePatternBrush

Api: Declare Function CreatePatternBrush Lib "gdi32" (ByVal hBitmap As Long) As Long

Aquí pueden ver que el único parámetro es un BITMAP.

OTRAS API’S

Seguramente vieron sencilla la explicación de las 3 API ‘S que se usan para crear una brocha pero de seguro no entienden todavía el programa, ya que hay mas API’S en juego que de seguro los confunde, básicamente para crear una brocha se necesita las API que ya mencione y las que voy a explicar ahora son algunas API’S adicionales y otras API’S obligadas para el funcionamiento de la brocha.

Pero antes de entrar en detalle necesito explicar una cosita, verán que hay dos API SelectObject y DeleteObject.

Cuando uno crea un HDC el viene con un paquete incluido, ese paquete tiene brocha, un pincel, una fuente, etc. En este momento cubriremos nada más la brocha. Es algo así:

Noten que en las API’S que crean las brochas no se menciona el HDC en ningún lado, la misión de esas API’S es simple: ¡crear algo! , en este caso una brocha. Pero es la función SelectObJect la encargada de cambiar la brocha existente por una nueva.

SelectObject

Api: Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Vean que aquí sí hay un hdc, que es nuestro primer parámetro y hay una variable que es nuestro objeto. Analicemos una de las líneas de código del programa:

        brocha = CreateHatchBrush(HS_CROSS, RGB(0, 0, 0))

        SelectObject Me.hdc, brocha

Vean que primero creamos la brocha, la cual almacenamos en una variable “brocha”, pero esa brocha esta en el aire, esta en memoria, no pertenece a nadie, y luego SelectObject entra en juego, lo que hace en “español” es lo siguiente: “Cámbiame la brocha que tiene Me.hdc, por esta nueva que se llama brocha”. Ustedes dirán: pero si Me.hdc no tiene brocha ya que yo no configure ninguna, ¡error!,  Como dije arriba todo HDC viene con un paquete y ese paquete contiene una brocha.

Ahora ¿se pueden hacer mas cosas con SelectObject?.si!. Que sucede si hago esto:

        brocha = CreateHatchBrush(HS_CROSS, RGB(0, 0, 0))

        brochavieja = SelectObject(Me.hdc, brocha)

Se ve como obvio, si ustedes tomaran el valor de retorno de SelectObject, no sería más que el objeto que fue suplantado, que quiero decir, recuerdan que Me.hdc ya tiene una brocha, y con SelectObject la cambian, el valor que retorna SelectObject es la brocha antigua que tenia nuestro HDC. ¿Para que quiero la brocha antigua?. No se!, dependerá del programa que hagas pero algunas veces suele ser útil, ¿Cómo retorno la brocha antigua? por ejemplo:

      
brocha = CreateHatchBrush(HS_CROSS, RGB(0, 0, 0))

      
brochavieja = SelectObject(Me.hdc, brocha)

Rectangle Me.hdc, 0, 0, 100, 100


SelectObject Me.hdc, brochavieja

Rectangle Me.hdc, 0, 0, 100, 100

Lo que estamos haciendo es pintar un rectángulo con una brocha nueva, en este caso de tramado, luego suplantamos o cambiamos la brocha nueva por la vieja, y pintamos un rectángulo con la brocha vieja. Pregunto algo: ¿si yo hiciera esto en el código de arriba?:


BrochaX = SelectObject(Me.hdc, brochavieja)

Pregunto ¿Que brocha tiene brochaX?. Después de todo de lo que he explicado deberían de conocer la respuesta por ende no se las diré.

Esta parte no la tengo en el código de ejemplo, por que no era necesario almacenar la brocha antigua, ya que no me interesaba, pero uno nunca sabe los programas que tenemos que crear.

Si creamos un objeto en memoria, ese objeto lo tenemos que eliminar, como las API es independiente de VB si ustedes crean un objeto con las funciones API, el hecho que cierren el programa no implica que VB quitara de memoria ese objeto, y para ello usamos DeleteObject.

DeleteObject

Api: Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Fácil de explicar, lo que hace es eliminar el objeto creado. En nuestro código en Form_Unload tenemos lo siguiente:

    DeleteObject brocha

Lo que hacemos aquí es eliminar nuestra brocha creada. Para que no quede en la memoria. Pero ustedes dirán: “Eduardo pero como sabemos que API’S usan SelectObject para su funcionamiento”, básicamente en la documentación les dice, pero aquí les menciono algunas funciones que usan SelectObject en su funcionamiento, y evidentemente si usan SelectObject deberían de usar DeleteObject para borrarlo, aquí están las funciones:

CreateBitmap, CreateBitmapIndirect, CreateCompatibleBitmap, CreateDIBitmap, CreateDIBSection, CreateBrushIndirect, CreateDIBPatternBrush, CreateDIBPatternBrushPt, CreateHatchBrush, CreatePatternBrush, CreateSolidBrush, CreateFont, CreateFontIndirect, CreatePen, CreatePenIndirect, CombineRgn, CreateEllipticRgn, CreateEllipticRgnIndirect, CreatePolygonRgn, CreateRectRgn, CreateRectRgnIndirect

Aunque en este manual no trabajaremos todas estas API, es importante que sepan cuando tienen que usar SelectObject.

Sigamos con las API’S extrañas que están en el código. Nos faltan 3 Api’s que la explicamos a continuación: 

GetStockObject

Api: Declare Function GetStockObject Lib "gdi32" (ByVal nIndex As Long) As Long

Windows como sistema operativo tiene objetos predefinidos, es decir, objetos ya creados y la forma de invocarlos o usarlos es con esta función GetStockObject, su único parámetro es un índice que dependiendo del índice que sea, retornara un objeto especifico.

Los posibles nIndex son:

Private Const BLACK_BRUSH = 4  ‘Brocha de color NEGRO
Private Const DKGRAY_BRUSH = 3 ‘Brocha de color Gris Oscuro
Private Const GRAY_BRUSH = 2 ‘Brocha de color Gris
Private Const HOLLOW_BRUSH = 5 ‘Brocha Nula o vacia
Private Const LTGRAY_BRUSH = 1 ‘Brocha gris claro
Private Const WHITE_BRUSH = 0 ‘Brocha blanca
Private Const WHITE_PEN = 6 ‘Pincel blanco
Private Const ANSI_FIXED_FONT = 11 ‘Fuente utilizada por el sistema
Private Const ANSI_VAR_FONT = 12 ‘Igual que la anterior, pero con paso de variable.
Private Const DEFAULT_GUI_FONT = 17 ‘Fuente por defecto en Windows utilizada en los menús y cajas de diálogos
Private Const SYSTEM_FONT = 13 ‘Fuente del sistema
Private Const SYSTEM_FIXED_FONT = 16 Esta permite compatibilidad con window 3.1 o 16 bits.
Private Const DEFAULT_PALETTE = 15 ‘Paleta por defecto, usada en la creación de imágenes.
Estas últimas variables poco usadas se usan cuando se crean fuentes.

¿Cómo usamos esta función un nuestro código?. Veamos:

SelectObject Me.hdc, GetStockObject(LTGRAY_BRUSH)

Rectangle Me.hdc, 0, 0, 100, 100

Observen que en ningún momento uso ninguna función Create…Brush, ya que la función GetStockObject me retorna una brocha del sistema o ya configurada en Windows, por lo que el trabajo se hace directo.

GetSysColor

Api: Declare Function GetSysColor Lib "user32" (ByVal nIndex As Long) As Long

Así como Windows posee objetos predefinidos o creados, también Windows posee una configuración de los colores del sistema. Que quiero decir, si tuvieran Microsoft Plus, notaran que cuando cambia las opciones ven diferentes colores, en los botones, formularios, caption, etc. Windows mantiene un registro de esos colores en el registro de Windows y la forma de acceder a esos colores es por esta función, con esto lo que se quiere es que las aplicaciones sean compatible con el color que define el cliente. Esta API es sencilla lo que hace es regresar el color correspondiente a un índice. Los cuales presento a continuación:

Valores posibles para “nIndex”:

Private Const COLOR_3DDKSHADOW = 21 ‘Color Sombra oscura de los elementos tridimensionales
Private Const COLOR_3DLIGHT = 22 ‘Color de la cara de los elementos tridimensionales
Private Const COLOR_BTNFACE = 15 ‘El mismo color que el anterior pero de los botones
Private Const COLOR_BTNHIGHLIGHT = 20 ‘Color brillante de los elementos tridimensionales
Private Const COLOR_BTNSHADOW = 16 ‘Color Sombra de los elementos tridimensionales
Private Const COLOR_ACTIVEBORDER = 10 ‘Color del borde activo
Private Const COLOR_ACTIVECAPTION = 2 ‘Color de la barra de titulo
Private Const COLOR_BACKGROUND = 1 ‘Color del escritorio
Private Const COLOR_BTNTEXT = 18 ‘Color del texto en los botones
Private Const COLOR_CAPTIONTEXT = 9 ‘Color del texto en las etiquetas
Private Const COLOR_GRADIENTACTIVECAPTION = 27 ‘Color del lado izquierdo de la barra de titulo cuando esta activo.
Private Const COLOR_GRADIENTINACTIVECAPTION = 28 ‘Color del lado izquierdo de la barra de titulo cuando esta inactivo. Estas ultimas 2 son nada mas para NT 5.0 y Windows 98, si ven la barra superior del formulario verán un degradado, de un color a otro, bueno ambas variables toman el color del lado izquierdo el primero es cuando esta activo y el segundo cuando esta inactivo.
Private Const COLOR_GRAYTEXT = 17 ‘Color del texto cuando esta desactivado
Private Const COLOR_HIGHLIGHT = 13 ‘Color de los ítem’s seleccionados de un control.
Private Const COLOR_HIGHLIGHTTEXT = 14 ‘Color del texto cuando es seleccionado de un control
Private Const COLOR_INACTIVEBORDER = 11 ‘Color del borde inactivo
Private Const COLOR_INACTIVECAPTION = 3 ‘Color del caption inactivo
Private Const COLOR_INACTIVECAPTIONTEXT = 19 ‘Color del texto inactivo
Private Const COLOR_INFOBK = 24 ‘Color del fondo de los cuadritos ToolTip
Private Const COLOR_INFOTEXT = 23 ‘Color del texto de los cuadritos ToolTip
Private Const COLOR_MENU = 4 ‘Color del fondo del menú

Private Const COLOR_MENUTEXT = 4 ‘Color del texto del menú
Private Const COLOR_SCROLLBAR = 0 ‘Color del scroll bar
Private Const COLOR_WINDOW = 5 ‘Color del fondo de window (formulario)
Private Const COLOR_WINDOWFRAME = 6  ‘Color del frame de window (formulario)
Private Const COLOR_WINDOWTEXT = 8  ‘Color del texto de window (formulario)
Y si revisan la documentación oficial en Internet pueden encontrar un poquito mas. Pero básicamente esto son todos los colores del sistema.

El uso de esta API se hace de esta manera:

brocha = CreateSolidBrush(GetSysColor(COLOR_SCROLLBAR))

SelectObject Me.hdc, brocha

Creo que es fácil ver que hace, GetSysColor nos regresa un color dependiendo de las constante que le pasamos.

La última API Rectangle, no la explico ya que tengo una sección para ella, pero creo que es muy fácil de entender.

Esto no es todo sobra brochas, hay mas API’S pero para mi estas son las mas importante y las mas usadas. De todas maneras en la documentación oficial en Internet encontraran todos las API’S Win32 referente a las brochas.

PINCELES

Muy bien, ya vimos las brochas, ahora nos toca los pinceles pero, ¿Qué es el pincel?  si volvemos al pintor, pincel es lo que se usa para pintar las líneas de una figura, claro que no todos están de acuerdo conmigo, ya que con una brocha en la vida real ustedes puede pintar líneas, eso es cierto, pero en la computadora el pincel es el encargado del pintado de las líneas y las brochas es en el encargado del relleno. Es como si el pintor usara un pincel fino para delimitar un triangulo y luego usa un pincel grueso “brocha” para rellenar el triangulo de un color.

Ahora en la API Rectangle, pudimos ver como se llenaba el fondo con un color pero si notan bien había un marco que delimitaba al rectángulo, trabajar con ese marco es trabajar con el pincel, para cualquier figura que veamos en el futuro, veremos que todas las figuras están delimitadas por líneas, y usamos los pinceles para alterar o modificar esas líneas, para obtener un efecto cualquiera.

Windows trabaja con 2 tipos de pinceles: Cosméticos y Geométricos.

Los cosméticos son pinceles que se dibujan rápidamente. Estos pinceles se dibujan directamente en el dispositivo, sin pasar por el motor de transformación de Windows. Es decir, sus unidades son de dispositivo y no lógicas.

En cambio, los Geométricos son pinceles un poco lentos, pero hacen un uso completo del motor de transformación de Windows. Por lo que sus unidades son lógicas y no físicas. Ustedes dirán,  son malos estos pinceles, no!, son mejores, tardan mas pero se pueden lograr efectos mucho mas avanzados que con los pinceles cosméticos. Lo único lamentable es que dichos efectos tienen resultado en su mayoría en Windows 2000/XP/NT.

En este momento ustedes están confundidos ya que les mencione “motor de transformación de Windows”, “unidades lógicas”, “unidades físicas”, ya se que talvez esto no lo entienden ahora, pero aguanten un rato que después de los pinceles explicare esto. Algunos libros explican esta parte primero antes de explicar las herramientas graficas, particularmente yo prefiero explicar brochas y pinceles primero ya que necesito que vayan entrando a este mundo de las API, y mas que cuando explique esa parte utilizare lo ya explicado. Así que no importa si no entienden ahorita, después verán que todo era simple.

Pongan este programa en VB (para este proyecto necesitan 2 combobox,  debajo del Combo1 pongan un TextBox llamado Text1 y pónganle un valor por defecto que sea 1), y coloquen la imagen “pens.bmp” en la carpeta del proyecto:

Private Type LOGBRUSH

        lbStyle As Long

        lbColor As Long

        lbHatch As Long

End Type

Private Declare Function CreatePen Lib "gdi32" (ByVal nPenStyle As Long, ByVal nWidth As Long, ByVal crColor As Long) As Long

Private Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Declare Function ExtCreatePen Lib "gdi32" (ByVal dwPenStyle As Long, ByVal dwWidth As Long, lplb As LOGBRUSH, ByVal dwStyleCount As Long, lpStyle As Long) As Long

Private Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal dwImageType As Long, ByVal dwDesiredWidth As Long, ByVal dwDesiredHeight As Long, ByVal dwFlags As Long) As Long

Const LR_LOADFROMFILE = &H10

Const IMAGE_BITMAP = 0

Const IMAGE_ICON = 1

Const IMAGE_CURSOR = 2

Const IMAGE_ENHMETAFILE = 3

Const CF_BITMAP = 2

Private Const BS_SOLID = 0

Private Const BS_HATCHED = 2

Private Const BS_PATTERN = 3

Private Const HS_BDIAGONAL = 3               '  /////

Private Const HS_CROSS = 4                   '  +++++

Private Const HS_DIAGCROSS = 5               '  xxxxx

Private Const HS_FDIAGONAL = 2               '  \\\\\

Private Const HS_HORIZONTAL = 0              '  -----

Private Const HS_VERTICAL = 1                '  |||||

Private Const PS_GEOMETRIC = &H10000

Private Const PS_COSMETIC = &H0

Private Const PS_DASH = 1                    '  -------

Private Const PS_DASHDOT = 3                 '  _._._._

Private Const PS_DASHDOTDOT = 4              '  _.._.._

Private Const PS_DOT = 2                     '  .......

Private Const PS_SOLID = 0

Private Const PS_NULL = 5

Private Const PS_ENDCAP_FLAT = &H200

Private Const PS_ENDCAP_SQUARE = &H100

Private Const PS_ENDCAP_ROUND = &H0

Private Const PS_JOIN_BEVEL = &H1000

Private Const PS_JOIN_MITER = &H2000

Private Const PS_JOIN_ROUND = &H0

Private Sub Combo1_Click()

    Me.Cls

    Select Case Combo1.ListIndex

    Case 0

        nuevopen = CreatePen(PS_SOLID, Text1.Text, RGB(255, 0, 255))

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 10, 100, 100

        DeleteObject nuevopen

    Case 1

        nuevopen = CreatePen(PS_DASH, Text1.Text, RGB(255, 0, 255))

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 10, 100, 100

        DeleteObject nuevopen

    Case 2

        nuevopen = CreatePen(PS_DASHDOT, Text1.Text, RGB(255, 0, 255))

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 10, 100, 100

        DeleteObject nuevopen

    Case 3

        nuevopen = CreatePen(PS_DASHDOTDOT, Text1.Text, RGB(255, 0, 255))

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 10, 100, 100

        DeleteObject nuevopen

    Case 4

        nuevopen = CreatePen(PS_DOT, Text1.Text, RGB(255, 0, 255))

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 10, 100, 100

        DeleteObject nuevopen

    Case 5

        nuevopen = CreatePen(PS_NULL, Text1.Text, RGB(255, 0, 255))

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 10, 100, 100

        DeleteObject nuevopen

    End Select

End Sub

Private Sub Combo2_Click()

    Me.Cls

    Dim nuevopen As Long

    Dim nuevabrocha As LOGBRUSH

    Select Case Combo2.ListIndex

    Case 0

       nuevabrocha.lbColor = RGB(0, 0, 0)

       nuevabrocha.lbHatch = 0

       nuevabrocha.lbStyle = BS_SOLID       

       nuevopen = ExtCreatePen(PS_GEOMETRIC Or PS_ENDCAP_ROUND Or PS_JOIN_BEVEL, 10, nuevabrocha, 0, ByVal 0&)

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 40, 100, 140

        DeleteObject nuevopen

    Case 1

       nuevabrocha.lbColor = RGB(0, 255, 255)

       nuevabrocha.lbHatch = 0

       nuevabrocha.lbStyle = BS_SOLID

       nuevopen = ExtCreatePen(PS_GEOMETRIC Or PS_ENDCAP_ROUND Or PS_JOIN_MITER, 10, nuevabrocha, 0, ByVal 0&)

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 40, 100, 140

        DeleteObject nuevopen

    Case 2

        nuevabrocha.lbColor = RGB(255, 0, 255)

        nuevabrocha.lbHatch = 0

        nuevabrocha.lbStyle = BS_SOLID

        nuevopen = ExtCreatePen(PS_GEOMETRIC Or PS_DOT Or PS_ENDCAP_ROUND Or PS_JOIN_ROUND, 5, nuevabrocha, 0, ByVal 0&)

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 40, 100, 140

        DeleteObject nuevopen

    Case 3

        nuevabrocha.lbColor = RGB(0, 0, 0)

        nuevabrocha.lbHatch = LoadImage(App.hInstance, App.Path & "\pens.bmp", IMAGE_BITMAP, 8, 8, LR_LOADFROMFILE)

        nuevabrocha.lbStyle = BS_PATTERN

        nuevopen = ExtCreatePen(PS_GEOMETRIC, 25, nuevabrocha, 0, ByVal 0&)

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 40, 100, 140

        DeleteObject nuevopen

    End Select

End Sub

Private Sub Form_Load()

    Combo1.AddItem "PS_SOLID"

    Combo1.AddItem "PS_DASH"

    Combo1.AddItem "PS_DASHDOT"

    Combo1.AddItem "PS_DASHDOTDOT"

    Combo1.AddItem "PS_DOT"

    Combo1.AddItem "PS_NULL"

    Combo2.AddItem "PS_GEOMETRIC - PS_ENDCAP_ROUND - PS_JOIN_BEVEL"

    Combo2.AddItem "PS_GEOMETRIC - PS_ENDCAP_ROUND - PS_JOIN_MITER"

    Combo2.AddItem "PS_GEOMETRIC - PS_DOT - PS_ENDCAP_ROUND - PS_JOIN_ROUND"

    Combo2.AddItem "PS_GEOMETRIC - BS_PATTERN"

End Sub
Como se puede ver, el combo1 se encarga de los pinceles Cosméticos y el combo2 se encarga de los Geométricos. En caso de tener Windows 98/95/ME, no verán ningún efecto en el combo2. Verán un rectángulo con colores pero en Windows 2000/NT/XP se ve distinto.

CreatePen
Api: Declare Function CreatePen Lib "gdi32" (ByVal nPenStyle As Long, ByVal nWidth As Long, ByVal crColor As Long) As Long

Api sencilla encargada de crear un nuevo pincel “cosmético”. Como primer parámetro se pasa “nPenStyle” los cuales puede poseer los siguientes valores:

Private Const PS_DASH = 1                    '  -------

Private Const PS_DASHDOT = 3                 '  _._._._

Private Const PS_DASHDOTDOT = 4              '  _.._.._

Private Const PS_DOT = 2                     '  .......

Private Const PS_SOLID = 0

Private Const PS_NULL = 5

Estos valores se usan para configurar que tipo de trazado queremos. El parámetro “nWidth” se encarga de establecer el ancho del pincel, este parámetro solo sirve para el estilo PS_SOLID, en caso de no ser PS_SOLID, no importa el número que coloquen siempre será 1. Por último tenemos “crColor” encargado de determinar el color del pincel.

Creo que es fácil de ver lo que hace si paseamos por los distintos valores del combo1. Nada más explicare uno de ellos,  que vale para todos.

        nuevopen = CreatePen(PS_SOLID, Text1.Text, RGB(255, 0, 255))

        SelectObject Me.hdc, nuevopen

        Rectangle Me.hdc, 10, 10, 100, 100

        DeleteObject nuevopen

Al igual que la brocha, CreatePen nos retorna un valor el cual le pasamos a nuestro HDC por medio de SelecObject.

        nuevopen = CreatePen(PS_SOLID, Text1.Text, RGB(255, 0, 255))

Aquí estamos configurando un pincel SÓLIDO, cuya anchura es el valor que pongan en Text1, y cuyo color es morado. Como pueden ver se usa prácticamente de la misma manera que una brocha con la diferencia que esta altera el marco del rectángulo y no el fondo del mismo como lo hace la brocha. Prueben cambiando la anchura en PS_SOLID para que vean el efecto.

ExtCreatePen

Api: Declare Function ExtCreatePen Lib "gdi32" (ByVal dwPenStyle As Long, ByVal dwWidth As Long, lplb As LOGBRUSH, ByVal dwStyleCount As Long, lpStyle As Long) As Long

Una API mucho mas completa que permite la creación tanto de pinceles “cosméticos” como “geométricos”, en este código use esta API para los pinceles Geométricos, pero cuando explique los parámetros verán como usarla para los cosméticos.

Primer parámetro:

DwPenStyle: esta variable es mucho más compleja en esta función API. Ya que se divide en cuatro grupos.

Grupo 1 (Tipo de Pincel):


Private Const PS_GEOMETRIC = &H10000

Private Const PS_COSMETIC = &H0

Grupo 2 (Estilo del Trazado)


Private Const PS_DASH = 1                    '  -------

Private Const PS_DASHDOT = 3                 '  _._._._

Private Const PS_DASHDOTDOT = 4              '  _.._.._

Private Const PS_DOT = 2                     '  .......

Private Const PS_SOLID = 0


Private Const PS_ALTERNATE = 8

Grupo 3 (Acabado).


Private Const PS_ENDCAP_FLAT = &H200

Private Const PS_ENDCAP_SQUARE = &H100

Private Const PS_ENDCAP_ROUND = &H0

Imágenes tomadas de la ayuda de Microsoft MSDN en internet.

Grupo 4 (Enlace entre trazados).


Private Const PS_JOIN_BEVEL = &H1000

Private Const PS_JOIN_MITER = &H2000

Private Const PS_JOIN_ROUND = &H0

Imágenes tomadas de la ayuda de Microsoft MSDN en internet.
DwWidth: este parámetro controla el ancho del pincel. Cabe destacar que este ancho en los pinceles cosméticos cumple las mismas restricciones que en el API CreatePen.

Lplb: es una estructura para configurar una brocha para un pincel, esta brocha surte efecto cuando el ancho “dwWidth” es un número relativamente alto. Lo que hace es pintar la parte rellena del pincel con la brocha que se seleccione. Esta estructura es de este tipo:

Private Type LOGBRUSH

        lbStyle As Long

        lbColor As Long

        lbHatch As Long

End Type

LbStyle: se encarga de configurar el tipo de brocha:

Private Const BS_SOLID = 0 ‘ Color SÓLIDO
Private Const BS_HATCHED = 2 ‘ Tramado de la brocha, son las mismas que las vistas anteriores.
Private Const BS_PATTERN = 3 ‘ Utilizada en caso que se desea un bitmap como tramado
lbColor : determina el color de la brocha.

LbHatch: determina el tipo de tramado. Puede ser:

Imágenes tomadas de la ayuda de Microsoft MSDN en internet.

Sus constantes son:

Private Const HS_BDIAGONAL = 3               '  /////

Private Const HS_CROSS = 4                   '  +++++

Private Const HS_DIAGCROSS = 5               '  xxxxx

Private Const HS_FDIAGONAL = 2               '  \\\\\

Private Const HS_HORIZONTAL = 0              '  -----

Private Const HS_VERTICAL = 1                '  |||||

Queda claro que si “lbStyle” es BS_SOLID no importa que valor pongan en lbHatch, la API no lo tomara en cuenta.

Regresando con los parámetros de la API  ExtCreatePen, tenemos:

dwStyleCount y lpStyle no son tratadas en este manual y son utilizadas para pinceles personalizados. Si quieren más información diríjanse a la pagina de MSDN en Internet.

Ustedes dirán: en el primer parámetro ¿como combino todas esas constantes?. Algunas API’S win32 se le puede pasar como parámetro varias constantes, para ello usamos el operador lógico OR. Es decir, si queremos un pincel Geométrico, Sólido, con ENDCAP_ROUND, y Join_Bevel sería algo así:

ExtCreatePen( PS_GEOMETRIC OR PS_SOLID OR PS_ENCAP_ROUND OR PS_JOIN_LEVEL, ……..)

Esta sería la manera de pasarle 4 constantes en un solo parámetro. Cada API win32 en su documentación especifica si la API puede con esta posibilidad.

Pueden notar que es el mismo principio que las otras, con cierta diferencia. Es evidente que las combinaciones que se pueden lograr con el primer parámetro son muchas y no cubro todas sino 4 de ellas. Ustedes pueden jugar con ellas como quieran. Claro esta y creo que es como obvio que no se podría usar por ejemplo:

PS_JOIN_LEVEL OR PS_JOIN_MITER esto es por que o es un estilo o es el otro, pero no puedes mezclar dos estilos del mismo grupo.

¿Que funciona en Windows 98 y que funciona en Windows 2000/NT/XP?.-

Según la documentación encontre lo siguiente:

· El grupo 1 sirve para todos los Windows.

· El grupo 2 tiene lo siguiente:

· PS_DASH: en Windows 95 no funciona para pinceles geométricos. En Windows 98/ME: no funciona.

· PS_DOT: en Windows 95/98/ME no funciona con pinceles geométricos.

· PS_DASHDOT y PS_DASHDOTDOT: la misma restricción que PS_DASH.
· El grupo 3 y 4 solo funciona para pinceles geométricos.

Ahora todo esto se le dejo para quien lo quiera examinar minuciosamente, algunas condiciones no me convencen. Ya que como pueden ver en el combo2, ninguna selección funciona en Windows 98/95/ME. Solo vi el efecto en Windows 2000 y supongo que también funcionara en XP y NT.

Pero solo me remito a especificar lo que dice la documentación de Windows, puede que sea yo el que no sabe usar muy bien los parámetros. 

SISTEMA DE COORDENADAS Y MAPEADO
En esta sección necesito de su concentración ya que lo que voy a tratar es algo muy bonito e interesante pero el entenderlo implica un poco de esfuerzo por parte de nosotros, no tanto por las funciones API’S a tratar, sino que para hacerlas funcionar en VB tuve que valerme de un truco que estoy seguro que para la mayoría de ustedes será algo nuevo, así que ¡empecemos!.

Arriba mencione algo llamado “unidades lógicas” y “unidades físicas”, pero para llegar a que significa esto, vayamos a nuestro dispositivo de contexto HDC. Ustedes se han preguntado ¿Cuándo dibujábamos con la función RECTANGLE las coordenadas de los puntos, son píxeles?. Bueno la respuesta es si. Pero Windows permite dibujar en otro tipo de unidades, por ejemplo, podemos diseñar un rectángulo cuya medida en ves de ser píxel, puede ser milímetro o pulgadas. Es decir, veamos esta función:

        Rectangle Me.hdc, 10, 10, 100, 100

Lo que quiero decir arriba es que las coordenadas 10,10,100,100 pueden ser píxeles, pulgadas, milímetros, o una escala personalizada. En este momento de seguro mas de uno esta sorprendido, ya que creían que si querían hacer algo por el estilo ustedes pensaron que tenían que programar el código de conversión, pues debo decirles que Windows posee lo que se llama un “motor de transformación”. Esto es curioso, ya que ustedes se pueden preguntar ¿Si yo tengo mis coordenadas en milímetros, como Windows sabe cuantos píxeles equivale cada milímetro, o como maneja todo eso?. Esto sería una pregunta interesante, ya que si se dan cuenta el monitor dibuja en unidades de PIXELES y no en milímetro, ni pulgadas ni nada por el estilo. Entonces es aquí donde Windows separa la cuestión en dos estructuras el VIEWPORT y WINDOW (Ventana). 

La ventana (window) se mide en “unidades lógicas” , cuando me refiero a unidades lógicas me refiero, a unidades que pueden ser distinta de píxeles, como milímetro, pulgada, etc, y el viewport es medido en unidades físicas, es decir, PIXELES en caso del monitor, pero una unidad física de la impresora puede ser medida de otra manera, aunque en nuestro ejemplos usamos píxeles, ya que a la final todo es enviado al monitor; lo que hace Windows es una proyección desde la ventana al viewport.

Esa proyección es transparente para nosotros, el como Windows lo hace no importa. El motor de transformación de Windows esta formado por 4 espacios que veremos a continuación:

1.Espacio Mundo: este espacio es utilizado para efectuar transformaciones geométricas al dispositivo de contexto (HDC) seleccionado. Estas transformaciones son rotación, traslación, escalado, distorsión, espejo. Lamentablemente este espacio solo esta disponible en Windows 2000/XP/NT, los demás sistemas no poseen este espacio de trabajo.

2.Espacio Pagina: en este espacio es donde establecemos los unidades lógicas a utilizar, o véanlo como el lugar donde dibujamos nuestras figuras utilizando las unidades lógicas.

3.Espacio dispositivo: en este espacio se prepara el dispositivo de contexto para ser visualizado en la pantalla.

4.Espacio dispositivo físico: nos referimos a la unidad física que maneja el hardware en cuestión, si hablamos del monitor pues será píxeles, se es una impresora sería otra medida, etc. Ya aquí se muestra al cliente el resultado final, que es en pocas palabras lo que el cliente ve.

¿Qué es esto de espacio?. Cuando dibujamos en Windows, ​él pasa por los espacios 2,3,4, lo que sucede es que ustedes no se dan cuenta por que eso transparente tanto para el usuario como para nosotros, aunque de aquí en adelante estudiaremos el espacio 1 y 2, ya que son los únicos espacios que se pueden alterar, ya que el 3 y el 4 dependen mas de hardware que de software.

¿Por qué el espacio 1 no esta predeterminado?. Por el simple hecho de que no todas las versiones de Windows lo soporta, por lo tanto hay que activarlo.

También es importante saber que por ejemplo en nuestra ventana no necesariamente se debe de configurar el píxel o punto 0,0 en la parte superior izquierda, a nivel de hardware siempre va ser el 0,0 la parte superior izquierda, pero a nivel lógico el sistema de coordenadas puede cambiar pero ya hablaremos de eso mas adelante.

Una representación grafica de lo que hace Windows es algo por el estilo:

Básicamente lo que hace es lo siguiente, toma la función GDI, la cual se le pasa parámetros en unidades lógicas, llega al motor de conversión, si esta activado el espacio mundo, aplica las transformaciones solicitadas, en nuestro ejemplo es una transformación de distorsión, luego pasamos del espacio mundo al de pagina, aquí la imagen puede crecer o empequeñecer dependiendo de la escala programada (noten que nuestro sistema de coordenada no es necesariamente el 0,0 en la esquina superior izquierda), ya en el espacio de dispositivo la imagen se prepara para ser presentada, vean que el sistema de coordenada cambia al sistema conocido, luego se aplica por último la escala de visualización, es decir, el tamaño real que vera el usuario en su monitor, impresora, plotter, etc, y ya al final se muestra al usuario la imagen obviamente en unidades de píxeles si hablamos del monitor. En caso de no estar activado el espacio mundo, todo el proceso empieza en el espacio pagina.

Cabe mencionar que este proceso puede ir en orden inverso, es decir del hardware, al dispositivo de contexto.

Todos deben de estar esperando el proyecto de ejemplo, para ver como funciona las API que trabajan en todo este proceso, pero antes de eso necesito explicar algo.

En este proyecto verán que necesitamos un modulo, ¿Por qué?. Como todo sabemos VB tiene un formulario ya configurado cuando abren un proyecto, ese formulario tiene eventos, pero ahora necesito que recuerden el código inicial hecho en C, se recuerdan de la función “GestordeMensajes”, que yo les dije que era nuestro oído, que todos los mensajes que eran enviados a nuestra aplicación pasaban por esa función, es evidente ver que el formulario de VB no gestiona los mensajes de esa manera (desde nuestro punto de vista), ya que internamente el formulario debe de tener una función que gestiona los mensajes, lo que sucede es que es transparente para nosotros.

¿A que quiero llegar?. Existe un mensaje llamado WM_PAINT, este mensaje es enviado a nuestra ventana cuando ella requiere ser repintada, ya sea por que se mueve, o por que una ventana tapo a la nuestra, por X situación, internamente el formulario de VB debe gestionar el repintado del formulario, cosa o procedimiento transparente para nosotros, el problema que me encontré al trabajar en VB con las funciones correspondiente a este tema, es que de alguna manera el como gestiona VB el mensaje WM_PAINT, hacia que no funcionaran correctamente las funciones API’S, lo que producía errores extraños a las cuales le tuve que buscar solución.

Una de la solución fue crear mi propio “oído”, es decir, una función que capturara el mensaje WM_PAINT, y el pintado del formulario ya no quedaba bajo la responsabilidad de VB, sino bajo mi responsabilidad, es decir, en este proyecto nosotros somos los encargado de procesar lo que el HDC del formulario va a mostrar. Ya que si dejaba que el gestor que coloca VB para ello controlara el pintado, los resultados con estas funciones no eran óptimos.

Por esa razón, en este proyecto veremos funciones extrañas que no tienen que ver con el tema que tratamos pero que son necesaria para crear nuestro propio “oído”.

Para verlo mas claro vayamos al código. Para este proyecto necesitan de un formulario, con un control ComboBox, llamado combo1, traten de poner mas o menos grande el formulario y el combo pónganlo en la parte derecha del mismo, y van a crear un modulo, llamado “modulo1”.

En el formulario ponen este código:

Private Sub Combo1_Click()

    Me.Refresh

End Sub

Private Sub Form_Load()

    Cambiar_Proc (Me.hwnd)

    Combo1.AddItem "MM_TEXT"

    Combo1.AddItem "MM_HIENGLISH"

    Combo1.AddItem "MM_HIMETRIC"

    Combo1.AddItem "MM_LOENGLISH"

    Combo1.AddItem "MM_LOMETRIC"

    Combo1.AddItem "MM_TWIPS"

    Combo1.AddItem "TRASLADAR"

    Combo1.AddItem "ESCALAR"

    Combo1.AddItem "ROTACION"

    Combo1.AddItem "DISTORSION"

    Combo1.AddItem "REFLEXION"    

    Combo1.AddItem "COMBINACION TRASLADAR - ROTACION"

    Combo1.AddItem "PERSONALIZADO"

End Sub

En el modulo va el código interesante:

Option Explicit

Private Type Size

        cx As Long

        cy As Long

End Type

Private Type POINTAPI

        x As Long

        y As Long

End Type

Public Type XFORM

        eM11 As Single

        eM12 As Single

        eM21 As Single

        eM22 As Single

        eDx As Single

        eDy As Single

End Type

Public Type RECT

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

Public Type PAINTSTRUCT

        hdc As Long

        fErase As Long

        rcPaint As RECT

        fRestore As Long

        fIncUpdate As Long

        rgbReserved(32) As Byte

End Type

Declare Function SetWorldTransform Lib "gdi32" (ByVal hdc As Long, lpXform As XFORM) As Long

Declare Function SetGraphicsMode Lib "gdi32" (ByVal hdc As Long, ByVal iMode As Long) As Long

Declare Function SetMapMode Lib "gdi32" (ByVal hdc As Long, ByVal nMapMode As Long) As Long

Declare Function SetWindowExtEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpSize As Size) As Long

Declare Function SetViewportExtEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpSize As Size) As Long

Declare Function SetViewportOrgEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpPoint As POINTAPI) As Long

Declare Function SetWindowOrgEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpPoint As POINTAPI) As Long

Declare Function CombineTransform Lib "gdi32" (lpxformResult As XFORM, lpxform1 As XFORM, lpxform2 As XFORM) As Long

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Declare Function BeginPaint Lib "user32" (ByVal hwnd As Long, lpPaint As PAINTSTRUCT) As Long

Declare Function EndPaint Lib "user32" (ByVal hwnd As Long, lpPaint As PAINTSTRUCT) As Long

Public Const GM_ADVANCED = 2

Const MM_ANISOTROPIC = 8

Const MM_HIENGLISH = 5

Const MM_HIMETRIC = 3

Const MM_ISOTROPIC = 7

Const MM_LOENGLISH = 4

Const MM_LOMETRIC = 2

Const MM_TEXT = 1

Const MM_TWIPS = 6

Public Const WM_MOUSEMOVE = &H200

Private Const WM_PAINT = &HF

Public Const GWL_WNDPROC = -4

Public nuevoproc As Long

Public viejoproc As Long

Public mensageForm As String

Public brocha As Long

Public hdc As Long

Dim previoMap As Long

Dim punto As POINTAPI

Dim matriz As XFORM

Dim matriz2 As XFORM

Dim matriz3 As XFORM

Dim modomapeo As Integer

Sub Cambiar_Proc(hwnd2 As Long)

nuevoproc = ObtenerLong(AddressOf MenuProc)

viejoproc = SetWindowLong(hwnd2, GWL_WNDPROC, nuevoproc)

End Sub

Function ObtenerLong(ByVal direcfunt As Long) As Long

    ObtenerLong = direcfunt

End Function

Private Function MenuProc(ByVal hwnd As Long, ByVal message As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    If message = WM_PAINT Then

        Dim p As PAINTSTRUCT

        hdc = BeginPaint(hwnd, p)

        brocha = CreateSolidBrush(RGB(255, 100, 10))

        SelectObject hdc, brocha

        SetMapMode hdc, MM_TEXT

        SetViewportOrgEx hdc, 0, 0, punto

        Rectangle hdc, -50, -50, Form1.Width, Form1.Height

        DeleteObject brocha

        brocha = CreateSolidBrush(RGB(0, 0, 0))

        SelectObject hdc, brocha

        Select Case Form1.Combo1.ListIndex

            Case 0

                SetMapMode hdc, MM_TEXT

                SetViewportOrgEx hdc, 10, 10, punto

                Rectangle hdc, 0, 0, 100, 100

            Case 1

                SetMapMode hdc, MM_HIENGLISH

                SetViewportOrgEx hdc, 50, 50, punto

                Rectangle hdc, 0, 0, 100, 100

            Case 2

                SetMapMode hdc, MM_HIMETRIC

                SetViewportOrgEx hdc, 50, 50, punto

                Rectangle hdc, 0, 0, 100, 100

            Case 3

                SetMapMode hdc, MM_LOENGLISH

                SetViewportOrgEx hdc, 50, 90, punto

                Rectangle hdc, 0, 0, 100, 100

            Case 4

                SetMapMode hdc, MM_LOMETRIC

                SetViewportOrgEx hdc, 50, 50, punto

                Rectangle hdc, 0, 0, 100, 100

            Case 5

                SetMapMode hdc, MM_TWIPS

                SetViewportOrgEx hdc, 50, 50, punto

                Rectangle hdc, 0, 0, 100, 100

            Case 6

                With matriz

                    .eDx = 10

                    .eDy = 10

                    .eM11 = 1

                    .eM22 = 1

                    .eM12 = 0

                    .eM21 = 0

                End With

                SetGraphicsMode hdc, GM_ADVANCED

                SetWorldTransform hdc, matriz

                Rectangle hdc, 0, 0, 100, 100

                identidad hdc

            Case 7

                With matriz

                    .eDx = 0

                    .eDy = 0

                    .eM11 = 2

                    .eM22 = 2

                    .eM12 = 0

                    .eM21 = 0

                End With

                SetGraphicsMode hdc, GM_ADVANCED

                SetWorldTransform hdc, matriz

                Rectangle hdc, 20, 20, 100, 100

                identidad hdc

            Case 8

                With matriz

                    .eDx = 0

                    .eDy = 0

                    .eM11 = 2

                    .eM22 = 2

                    .eM12 = 2

                    .eM21 = 1

                End With

                SetGraphicsMode hdc, GM_ADVANCED

                SetWorldTransform hdc, matriz

                Rectangle hdc, 20, 20, 100, 100

                identidad hdc

            Case 9

                With matriz

                    .eDx = 0

                    .eDy = 0

                    .eM11 = 1

                    .eM22 = 1

                    .eM12 = 2

                    .eM21 = 4

                End With

                SetGraphicsMode hdc, GM_ADVANCED

                SetWorldTransform hdc, matriz

                Rectangle hdc, 20, 20, 100, 100

                identidad hdc

             Case 10

                With matriz

                    .eDx = 0

                    .eDy = 0

                    .eM11 = -1

                    .eM22 = -1

                    .eM12 = 0

                    .eM21 = 0

                End With

                SetMapMode hdc, MM_HIENGLISH

                SetViewportOrgEx hdc, 50, 50, punto

                Rectangle hdc, 20, 20, 50, 50

                SetGraphicsMode hdc, GM_ADVANCED

                SetWorldTransform hdc, matriz

                Rectangle hdc, 20, 20, 50, 50

                identidad hdc

            Case 11

                With matriz2

                    .eDx = 10

                    .eDy = 10

                    .eM11 = 1

                    .eM22 = 1

                    .eM12 = 0

                    .eM21 = 0

                End With

                With matriz3

                    .eDx = 0

                    .eDy = 0

                    .eM11 = 2

                    .eM22 = 2

                    .eM12 = 2

                    .eM21 = 1

                End With

                CombineTransform matriz, matriz2, matriz3

                SetGraphicsMode hdc, GM_ADVANCED

                SetWorldTransform hdc, matriz

                Rectangle hdc, 0, 0, 100, 100

                identidad hdc

            Case 12

                SetMapMode hdc, MM_ISOTROPIC

                Dim oldsize1 As Size

                Dim oldsize2 As Size

                SetWindowExtEx hdc, 396, 197, oldsize1

                SetViewportExtEx hdc, 792, 394, oldsize2

                Rectangle hdc, 0, 0, 198, 98

                SetWindowExtEx hdc, oldsize1.cx, oldsize1.cy, oldsize1

                SetViewportExtEx hdc, oldsize2.cx, oldsize2.cy, oldsize2

        End Select

        DeleteObject brocha

        EndPaint hwnd, p

    Else

        MenuProc = CallWindowProc(viejoproc, hwnd, message, wParam, lParam)

    End If

End Function

Sub identidad(hdc As Long)

    With matriz

        .eDx = 0

        .eDy = 0

        .eM11 = 1

        .eM22 = 1

        .eM12 = 0

        .eM21 = 0

    End With

    SetWorldTransform hdc, matriz

End Sub

Las siguientes funciones están afuera del contexto que queremos explicar acerca del sistema de coordenadas y modo de mapeo: SetWindowLong, CallWindowProc, BeginPaint, EndPaint. La pregunta principal debe de ser ¿Cómo creamos nuestro oído?.

Cabe mencionar lo siguiente, ya explique con anterioridad que VB coloca una función que gestiona los mensajes que nosotros no vemos, en ves de eso VB nos presenta ciertos eventos en donde nosotros ponemos nuestro código, pero estos eventos los coloca como funciones separadas y no juntas como en realidad sucede.

Como explique cuando mostré el código en C en el primer Capítulo , toda ventana tiene una función que sirve como “oído” de la aplicación, en realidad es una función situada en memoria, lo que vamos hacer no es eliminar dicha función, sino crear otra función que sirva para “escuchar” y luego mediante un API cambiar nuestro “oído” viejo por el nuevo.

Para empezar vayamos a nuestro formulario, en la parte de Form_load y verán escrita esta línea:

Cambiar_Proc (Me.hwnd)

Ahora vayamos al modulo:

Sub Cambiar_Proc(hwnd2 As Long)

nuevoproc = ObtenerLong(AddressOf MenuProc)

viejoproc = SetWindowLong(hwnd2, GWL_WNDPROC, nuevoproc)

End Sub

¿Que hacemos con la variable “nuevoproc”?, lo que estamos haciendo es obtener una referencia de la ubicación en memoria de la función “Menuproc”, para ello sirve el comando AddressOf, la función ObtenerLong no es nada del otro mundo, es la función expresada a continuación:

Function ObtenerLong(ByVal direcfunt As Long) As Long

    ObtenerLong = direcfunt

End Function

Lo que hace es convertir en un tipo long el valor que obtiene AddressOf.

Vean que nuestra nueva función, se llama MenuProc que en el modulo esta declarada así:

Private Function MenuProc(ByVal hwnd As Long, ByVal message As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Noten que esta función sirve como “oído”, ya que la estructura se presta para ello, ya que tiene los parámetros mencionados en el segundo Capítulo ,  que son necesarios para hacer que nuestra aplicación “escuche”.

Ya con la ubicación de memoria o referencia almacenada en la variable “nuevoproc” utilizamos la función:

viejoproc = SetWindowLong(hwnd2, GWL_WNDPROC, nuevoproc)

con esta función cambiamos nuestro “oído” viejo por uno nuevo, vean que como primer parámetro tenemos el HWND de nuestra ventana, como segundo parámetro establecemos la acción que queremos hacer, en este caso GWL_WNDPROC el cual es utilizado para manipular la función de mensajes, y como tercer parámetro le pasamos la dirección de la nueva función. Es fácil darse cuenta que lo que retorna la función es la dirección de la función vieja, ósea, la que VB configura por defecto. Esto quiere decir que ahora nuestra aplicación tiene 2 funciones aptas para escuchar mensajes, pero solo una esta activada por defecto, que en nuestro caso es “nuevoproc”, que referencia a la función  “MenuProc”.

¿Cómo escuchamos?.

Recuerdan que yo dije que para este proyecto funcionara bien teníamos que controlar de manera independiente el mensaje WM_PAINT, bueno creo que es fácil ver, que el IF hace eso. Aquí lo que estoy haciendo es que SI MESSAGE = WM_PAINT (ENTONCES YO ME ENCARGO DE PINTAR) DE LO CONTRARIO (ELSE) QUE LA FUNCIÓN VIEJA SE ENCARGE DEL MENSAJE
    If message = WM_PAINT Then

        Dim p As PAINTSTRUCT

        hdc = BeginPaint(hwnd, p)


.


.


.

        EndPaint hwnd, p

    Else

        MenuProc = CallWindowProc(viejoproc, hwnd, message, wParam, lParam)

    End If

Presten atención a lo que esta en azul; esta función (MenuProc) es la encargada de “escuchar”, pero yo no quiero capturar todos los mensajes ¿Por que?. por que sería un fastidio, ya VB me tiene una función ya configurada con esas cosas, lo que hago es que si MESSAGE es diferente a WM_PAINT se ejecuta la línea en azul, esta línea que usa CallWindowProc, es una función que se encarga de llamar a otra función que esta en capacidad de recibir un mensaje, y le dejo a la función vieja el trabajo de todos los demás mensajes. Ya que pueden observar que entre sus parámetros esta la variable “viejoproc”. Aislando solamente el mensaje WM_PAINT. 

Vean en el código de arriba que hay dos funciones y una estructura, BEGINPAINT, ENDPAINT y la estructura PAINTSTRUCT. Cuando uno pinta por el mensaje WM_PAINT uno tiene que preparar a la ventana para ser pintada y la manera de prepararla es con BEGINPAINT, luego de haber pintado cerramos con ENDPAINT y así le decimos a la ventana que esta lista para dibujarse, la estructura PAINTSTRUCT solo sirve de carácter informativo.

Mas o menos espero que hayan entendido un poco lo que hice, yo no voy a explicar esas funciones API’S mencionadas arriba, todavía, ya que esas funciones tienen un apartado especial en otra sección, solo necesito que sepan por encima lo que hice ya que lo que realmente voy explicar de ahora en adelante, va ser el código que hay entre BEGINPAINT y ENDPAINT, ese código es lo referente al tópico que estamos estudiando. Las funciones restantes que me sirvieron como ayuda para lograr mi meta en este Capítulo  serán estudiadas en sus respectivo tópicos o secciones mas adelante.

ESPACIO MUNDO

Este espacio que solo puede ser utilizado en sistemas operativos Windows 2000/NT/XP, es utilizado para hacer transformaciones geométricas al HDC.

Cabe mencionar que estas transformaciones se aplican a través de matrices, que en nuestro caso será una estructura de tipo XFORM, la cual llenándola de manera apropiada realiza cierta transformación.

Translación: esta transformación es fácil de entender ya que nada mas consta de trasladar nuestro HDC especificando el desplazamiento en el eje x e y.

Matriz:






Estructura








eM11 = 1
eM22 = 1

eM12 = 0
eM21 = 0








eDx = dx
eDy = dy

Siendo dx y dy, el desplazamiento horizontal y vertical respectivamente.

Escalado: transformación utilizada para escalar una imagen, logrando efectos de ampliación y disminución del tamaño del HDC.

Matriz:






Estructura








eM11 = dx
eM22 = dy

eM12 = 0
eM21 = 0








eDx = 0

eDy = 0 

Siendo dx y dy, el factor de multiplicación en el eje x e y respectivamente.

Rotación: transformación utilizada para rotar el HDC. 

Matriz:






Estructura








eM11 = cos(
[image: image4.wmf]q

) eM22 = cos(
[image: image5.wmf]q

)
eM12 = sen(
[image: image6.wmf]q

)
eM21 = -sen(
[image: image7.wmf]q

)







eDx = 0

eDy = 0 

Siendo 
[image: image8.wmf]q

 el ángulo de giro.

Distorsión: transformación utilizada para distorsionar un HDC.

Matriz:






Estructura








eM11 = 1
eM22 = 1

eM12 = x
eM21 = y








eDx = 0

eDy = 0 

Siendo x e y , la distorsión unitaria en su respectivos ejes.

Espejo: transformación utilizada para reflejar.

Matriz:






Estructura








eM11 = rx
eM22 = ry

eM12 = 0
eM21 = 0








eDx = 0

eDy = 0 

Si rx = -1 y ry = 1 se hace un reflejo horizontal

Si rx = 1 y ry = -1 se hace un reflejo vertical.

Los únicos valores posibles para rx y ry son 1 y –1

Funciones involucradas:

SetGraphicsMode

Api: Declare Function SetGraphicsMode Lib "gdi32" (ByVal hdc As Long, ByVal iMode As Long) As Long

Esta función es la encargada de activar el espacio mundo, por defecto sin importar el sistema operativo que se use, las transformaciones empiezan desde el espacio pagina, si se desea aplicar las transformaciones mencionadas, hay que utilizar esta función.

Como primer parámetro se le pasa el dispositivo de contexto, y como segundo parámetro se le pasa la constante:

Const GM_ADVANCED = 2

Pueden darse cuenta que en las opciones desde el 6 hasta el 11 del SELECT CASE colocado en la función MenuProc, utilizo esta función para activar el espacio mundo, queda entendido que estas transformaciones no sirven en Window 98/95/ME.

SetWorldTransform

Api: Declare Function SetWorldTransform Lib "gdi32" (ByVal hdc As Long, lpXform As XFORM) As Long

Función utilizada para realizar las transformaciones geométricas mencionadas arriba, como parámetro tiene el HDC al cual se aplicara la transformación que se le pasa en la estructura “lpXForm” que es del tipo XFORM. Los valores de esta estructura deben estar acorde a los parámetros explicados arriba, según sea la transformación que se quiera aplicar.

En los ejemplos notaran que primero lleno la estructura “matriz” luego activo el espacio mundo con la función ya mencionada y luego realizo mi transformación, Ej.:

SetWorldTransform hdc, matriz

Después que aplicas tu transformación procedes a dibujar lo que quieras dibujar, en nuestro caso dibujo un rectángulo.

CombineTransform

Api: Declare Function CombineTransform Lib "gdi32" (lpxformResult As XFORM, lpxform1 As XFORM, lpxform2 As XFORM) As Long

Algunas veces queremos aplicar mas de una transformación al mismo tiempo, y para ello hay dos opciones o aplican álgebra para lograr una matriz transformada o utilizan la función que le da Windows que en este caso es CombineTransform.

Vean que tiene nada mas 3 parámetros del tipo XFORM, de los cuales el primero es la matriz resultante de combinar las transformaciones de la estructura “lpxForm1” y “lpxForm2”.

El ejemplo en el código esta en la opción 11 del SELECT CASE y es así:

CombineTransform matriz, matriz2, matriz3

Donde “matriz”, es la estructura que posee los valores ya transformados de la matrices “matriz2” y “matriz3”.

En mi caso combine la translación con rotación.

ESPACIO PAGINA


En el espacio de pagina, Windows se encarga de realizar las conversiones pertinentes para luego enviar la información al dispositivo, dependiendo del sistema de mapeo seleccionado.

¿Qué es sistema de mapeo?. Es como determinar que medidas se va usar para graficar, es intuitivo pensar que cuando dibujamos lo estamos haciendo con una medida la cual es píxel; pero Windows permite la utilización de otros tipos de medidas e inclusive permite la creación de medidas personalizadas.

Modos de Mapeo

Nombre

Unidades 

Características
Valor de la constante

MM_ANISOTROPIC
configurable (X!=Y)
Personalizada

MM_ANISOTROPIC = 8

MM_HIENGLISH
0,001 pulgadas

Independiente

MM_HIENGLISH = 5

MM_HIMETRIC

0,01 milímetros

Independiente

MM_HIMETRIC = 3

MM_ISOTROPIC
configurable (x=y)
Personalizable

MM_ISOTROPIC = 7

MM_LOENGLISH
0,01 pulgadas

Independiente

MM_LOENGLISH = 4

MM_LOMETRIC
0,1 milímetros

Independiente

MM_LOMETRIC = 2

MM_TEXT

1 píxel


Dependiente

MM_TEXT = 1

MM_TWIPS

1/1440 pulgadas
Independientes

MM_TWIPS = 6

Cuando  me refiero a “independiente” me refiero a que es independiente del dispositivo. Con esto podemos ver que cuando configuramos las coordenadas como MM_HIMETRIC, por ejemplo, las medidas que toman las funciones de dibujo son en milímetros, es decir, cada píxel equivale en este caso a 0,01 milímetro.

Antes de hablar de las funciones encargadas de este trabajo, vamos a hablar del sistema de coordenada,  nosotros estamos acostumbrados a que la coordenada 0,0 es la esquina superior izquierda, pero esto no tiene por que ser así, se puede cambiar para formar un eje como el que usamos en matemáticas, y me refiero al eje de los positivos y negativos, con el sistema de coordenadas por defecto tenemos nada mas a nuestra disposición los valores positivos y aunque los valores negativos están, no lo vemos ya que están fuera del área de visualización del usuario.

Algunos sistema de mapeo funcionan como el sistema tradicional, es decir, los positivos del eje X están a la derecha y los del eje Y están hacia arriba, y no como estamos acostumbrados donde el los positivos del eje Y están hacia abajo.

MM_HIMETRIC, MM_LOENGLISH, MM_LOMETRIC, MM_TWIPS, MM_HIENGLISH = el eje X se incrementa izquierda a derecha, y el eje Y de abajo hacia arriba, como el típico sistema matemático.

MM_TEXT = el eje X se incrementa izquierda a derecha, y el eje Y de arriba hacia abajo.

Esto lo menciono ya que primero voy explicar la función encargada de mover el sistema de coordenadas.

SetWindowOrgEx

Api: Declare Function SetWindowOrgEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpPoint As POINTAPI) As Long

SetViewportOrgEx

Api: Declare Function SetWindowOrgEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpPoint As POINTAPI) As Long

Estas funciones hacen exactamente lo mismo pero una  trabaja en el espacio de pagina (SetWindowOrgEx) y la otra funciona solo en el espacio de dispositivo (SetViewportOrgEx). Ahora ¿Qué es eso de ViewPort y Window? Aunque arriba explique un poco esto, no importa, lo vuelvo a explicar. Véanlo así, la ventana (window) no es mas que una proyección al Viewport, ¿Qué esto?. Ustedes trabajan en un HDC, el cual puede estar configurado con un sistema de coordenada distinto, al sistema de coordenada del viewport, el cual es la imagen final que ve el cliente en su pantalla, pueden tener un sistema de coordenada en el espacio pagina, con valores negativos, pero si dejan el viewport como esta por defecto, puede ser que lo que tengan en esa área, no se muestre ya que lo que él hace es una proyección de un lugar a otro, inclusive en el HDC la figura puede ser muy pequeña y en el momento de mostrarlo por pantalla se puede ver grande, todo depende de cómo se configure todo esto, pero eso le veremos mas adelante.

Estas funciones fijan el centro de nuestro sistema de coordenada en el valor X e Y especificado, el parámetro “lpPoint” sirve para almacenar el punto anterior donde estaba el sistema de coordenadas.

Vean que en nuestro código es así:

SetViewportOrgEx hdc, 10, 10, punto

Lo que hago aquí es mover el centro de nuestro sistema de coordenada que inicialmente esta en el 0,0 al  punto 10,10, es decir, que si en nuestro HDC se encuentra algo dibujado en el punto 0,0 cuando el se proyecte al espacio de dispositivo no va ir al punto 0,0, sino que empieza en el punto 10,10, ¿Lo ven?. Fíjense que yo no cambie el sistema de coordenada de nuestro HDC, nuestro HDC maneja el sistema tradicional 0,0 esquina superior izquierda, solo cambie el del espacio de dispositivo.

Yo se que suena todo como complicado, pero pónganse a cambiar valores, jueguen con todo esto y de seguro entenderán mucho mas.

OffsetViewportOrgEx

Api: Declare Function OffsetViewportOrgEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpPoint As POINTAPI) As Long

OffsetWindowOrgEx

Api: Declare Function OffsetWindowOrgEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpPoint As POINTAPI) As Long

Alguna veces no queremos mover el sistema de coordenada de un punto a otro de manera arbitraria, sino queremos desplazarlo, estas funciones tienen los mismos parámetros que las dos funciones mencionadas arriba con la diferencia que estas dos funciones desplazan el origen. Veamos un ejemplo:

Imaginemos que tenemos el origen en 10,10 y hago esta instrucción:

SetViewportOrgEx hdc, 20, 20, punto

En este momento nuestro origen esta en el punto 20,20, pero si hago esto:

OffsetViewportOrgEx hdc, 20, 20, punto

Nuestro origen esta en el punto (30,30), ¿Ven por que?. Esta función desplaza al origen dependiendo los parámetro que ponen en “nX” y “nY”, desplazar es como sumarle al punto actual.

Estas funciones no las uso en el programa pero es importante explicarlas.

SetMapMode

Api: Declare Function SetMapMode Lib "gdi32" (ByVal hdc As Long, ByVal nMapMode As Long) As Long

Función encargada de establecer el sistema de mapeo que se quiera usar para el HDC escogido, el cual se le pasa como primer parámetro. “nMapMode”: en el se le coloca la constante correspondiente de la tabla Modos de Mapeo la cual mencione arriba.

Viendo un ejemplo:

   SetMapMode hdc, MM_HIENGLISH

                SetViewportOrgEx hdc, 50, 50, punto

                Rectangle hdc, 0, 0, 100, 100

Aquí cambio nuestro sistema a pulgadas, es decir un píxel ahora valdrá 0,001 pulgadas, es esa la razón del por que ven cuadros con distintos tamaños. Vean que el API Rectangle en casi todos los ejemplos tiene como coordenadas (0,0,100,100). Aquí este 100 no se refiere a píxeles sino pulgadas.

Esta explicación vale para cada opción ¿Por qué moví el centro del eje de coordenadas?. Recuerdan que arriba les dije como se orientan los ejes dependiendo del modo de mapeo escogido, lo que sucede es que si ponen en comentario esta línea no verán el rectángulo ¿Por qué?, por que el 100, correspondiente al eje Y va orientado hacia arriba y no hacia abajo como estamos acostumbrados.

SetWindowExtEx

Api: Declare Function SetWindowExtEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpSize As Size) As Long

SetViewportExtEx

Api: Declare Function SetViewportExtEx Lib "gdi32" (ByVal hdc As Long, ByVal nX As Long, ByVal nY As Long, lpSize As Size) As Long

Cuando usamos el modo de mapeo MM_ISOTROPIC o MM_ANISOTROPIC, estamos diciéndole que vamos a personalizar las medidas de nuestro sistema de coordenadas, con SetWindowExtEx establecemos las medidas en nuestro HDC y con SetViewportExtEx establecemos las medidas del espacio de dispositivo.

Los parámetros son iguales a los anteriores así que me evito repetir lo mismo.

Para entender como es esto veamos el ejemplo del programa:

SetWindowExtEx hdc, 396, 197, oldsize1

SetViewportExtEx hdc, 792, 394, oldsize2

Rectangle hdc, 0, 0, 198, 98

Para entender los números, primero tengo que decir que el formulario donde hice el ejemplo tenia de ancho 12000 y de alto 6315, lo equivalente en píxel es 792 x 394, vean que al viewport le estoy asignando ese mismo tamaño, y que al HDC o espacio de pagina le asigno la mitad de los valores.

Si ustedes ponen el formulario del mismo tamaño verán el efecto, cuando arrancan y seleccionan esa opción verán un rectángulo que cubre la mitad de ancho del formulario y la mitad de alto.

¿POR QUÉ?. Si la mitad de 792 es 396 y la de 394 es 197, por que con las coordenadas 198, 98 se llego hasta la mitad, suena como ilógico, pero no lo es, lo que hice fue que cada píxel en el HDC equivale a 2 píxeles en el Viewport ¡GUAO! dirán!, pongan ahora en ves de 198, ponen 396 verán que se llena todo el ancho del formulario ¡GUAO! de nuevo,  ven lo que sucede, esto no es mas que relacionar un sistema de coordenadas del espacio de pagina con el sistema de coordenadas del espacio de dispositivo, con esto pueden lograr efectos muy bueno a nivel de ZOOM o en programas que requieran medidas personalizadas. Pare entender esto, pónganse a cambiar los valores.

A partir de ahora vamos a ver funciones que se utilizan  para dibujar, las coordenadas de esta funciones son en unidades lógicas, no menciono eso, ya que sería repetir lo mismo para cada función API por lo que aviso desde ahorita, tomen esto como una regla, todo función que tenga un HDC, las coordenadas que maneja vienen en unidades lógicas, “lógicas”significa que dependerá del modo de mapeo seleccionado o configurado. Hay API’S que son excepciones pero son muy pocas, de todas maneras cuando lean en la documentación oficial verán la palabra “unit logical” = “unidades lógicas” y ya sabrán de lo que significa.

NOTA: pueden que al modificar el código, se hayan encontrado que si se equivocaron en el nombre de una variable o escribieron mal algo, VB los saca de manera arbitraria del código, en Windows 98 he notado que me sale un error y me saca, y en Windows 2000 lo que hace es cerrarme el programa, si las personas no saben de lo que hablo, pongan un nombre de variable que no exista en el código. Al correr el programa se darán cuenta del error, en realidad no se por que la causa del mismo, he de suponer que ha VB no le gusta que cambie el Gestor de mensajes “oído”, por lo que encuentra un conflicto interno cuando quiere lanzar el error y se da cuenta que el gestor de mensajes es otro. Pero esto es un detalle que no implica mal funcionamiento. Inclusive que esto no les de miedo de hacer una aplicación usando estas funciones, mientras programen bien, quitando todos los errores posibles no habrá problema. Ahora si alguien pudo codificar estas transformaciones utilizando el formulario sin la necesidad de cambiar la función que gestiona los mensajes le agradecería que me dijera como lo hizo, ya que yo trate y no se por que no funcionaba. De todas maneras seguiré buscando la causa.

FIGURAS GEOMÉTRICAS

Ahora vamos a introducir algunas figuras geométricas que posee Windows, para este proyecto solo necesito un formulario y un ComboBox llamado “Combo1”, traten de colocar el ComboBox pegado a la derecha del formulario.

Luego que tengan esto listo peguen este código en el formulario:

Private Declare Function Chord Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Private Declare Function Ellipse Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function FillRect Lib "user32" (ByVal hdc As Long, lpRect As rect, ByVal hBrush As Long) As Long

Private Declare Function FrameRect Lib "user32" (ByVal hdc As Long, lpRect As rect, ByVal hBrush As Long) As Long

Private Declare Function InvertRect Lib "user32" (ByVal hdc As Long, lpRect As rect) As Long

Private Declare Function Pie Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Private Declare Function Polygon Lib "gdi32" (ByVal hdc As Long, lpPoint As POINTAPI, ByVal nCount As Long) As Long

Private Declare Function PolyPolygon Lib "gdi32" (ByVal hdc As Long, lpPoint As POINTAPI, lpPolyCounts As Long, ByVal nCount As Long) As Long

Private Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function RoundRect Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long) As Long

Private Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Type POINTAPI

        X As Long

        Y As Long

End Type

Private Type rect

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

Dim brocha As Long

Private Sub Combo1_Click()

    Dim rect As rect

    rect.Left = 10

    rect.Right = 10

    rect.Bottom = 100

    rect.Right = 200

    Dim coordPoly(1 To 3) As POINTAPI

    Dim coordPoly2(1 To 6) As POINTAPI

    Dim vertPoly(0 To 1) As Long

   Select Case Combo1.ListIndex

    Case 0

        Me.Cls

        Chord Me.hdc, 10, 10, 200, 100, 105, 100, 0, 0

    Case 1

        Me.Cls

        Ellipse Me.hdc, 10, 10, 200, 100

    Case 2

        FillRect Me.hdc, rect, CreateSolidBrush(RGB(255, 255, 0))

    Case 3

        InvertRect Me.hdc, rect

    Case 4

        Me.Cls

        Pie Me.hdc, 10, 10, 100, 100, 10, 0, 50, 110

    Case 5

        Me.Cls

        FrameRect Me.hdc, rect, CreateSolidBrush(RGB(0, 0, 255))

    Case 6

        Me.Cls

        brocha = CreateSolidBrush(RGB(255, 0, 0))

        SelectObject Me.hdc, brocha

        coordPoly(1).X = 50

        coordPoly(1).Y = 10

        coordPoly(2).X = 25

        coordPoly(2).Y = 100

        coordPoly(3).X = 75

        coordPoly(3).Y = 100

        Polygon Me.hdc, coordPoly(1), 3

        DeleteObject brocha

    Case 7

        Me.Cls

        coordPoly2(1).X = 50

        coordPoly2(1).Y = 10

        coordPoly2(2).X = 25

        coordPoly2(2).Y = 100

        coordPoly2(3).X = 75

        coordPoly2(3).Y = 100

        vertPoly(0) = 3

        coordPoly2(4).X = 50

        coordPoly2(4).Y = 150

        coordPoly2(5).X = 50

        coordPoly2(5).Y = 80

        coordPoly2(6).X = 100

        coordPoly2(6).Y = 120

        'vertPoly(1) = 2

        PolyPolygon Me.hdc, coordPoly2(1), vertPoly(0), 1

    Case 8

        Me.Cls

        Rectangle Me.hdc, 10, 10, 200, 100

    Case 9

        Me.Cls

        brocha = CreateSolidBrush(RGB(255, 255, 100))

        SelectObject Me.hdc, brocha

        RoundRect Me.hdc, 10, 10, 200, 100, 40, 40

        DeleteObject brocha

    End Select

    Me.Refresh

End Sub

Private Sub Form_Load()

Me.AutoRedraw = True

Combo1.AddItem "Chord"

Combo1.AddItem "Ellipse"

Combo1.AddItem "Fillrect"

Combo1.AddItem "InvertRect"

Combo1.AddItem "Pie"

Combo1.AddItem "FrameRect"

Combo1.AddItem "Polygon"

Combo1.AddItem "PolyPolygon"

Combo1.AddItem "Rectangle"

Combo1.AddItem "RoundRect"

End Sub

Como ya habrán visto, al moverse por el ComboBox, el formulario visualiza cada figura, las cuales explicamos sus API’S a continuación (cabe destacar, que todas las API’S reciben entre sus parámetros el HDC o dispositivo de contexto en donde se dibujara la imagen):

Chord

Api: Declare Function Chord Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Para visualizar más claramente el papel que juega cada parámetro vean la figura:







Como pueden ver la figura habla por si sola, cabe destacar que no importa que vean un circulo, elipse, pie, etc, todo esas figuras se encuentran enmarcadas por un rectángulo. Los puntos X3,Y3, X4, Y4 son lo que determinan donde corta la recta a la elipse.

Ellipse

Api: Declare Function Ellipse Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Función API sencilla como pueden ver, lo único que hay que pasarle como parámetro son las coordenadas del rectángulo en la cual va estar encerrada la elipse.

FillRect

Api: Declare Function FillRect Lib "user32" (ByVal hdc As Long, lpRect As rect, ByVal hBrush As Long) As Long

Esta se encarga de mostrar un rectángulo con relleno, como vimos arriba para rellenar una figura necesitamos una brocha, la diferencia de esta API con las vistas anteriormente, es que la brocha no se configura a un dispositivo de contexto (HDC), sino a la figura misma, el hecho de que usemos una brocha tramada, de color, y texturizada para esta función, no implica que el HDC tome esa brocha. 

Las coordenadas del rectángulo son pasadas por la estructura de tipo RECT.

InvertRect
Api: Declare Function InvertRect Lib "user32" (ByVal hdc As Long, lpRect As rect) As Long

En FILLRECT vieron un rectángulo amarillo, y InvertRect observaron un rectángulo azul, lo que hace InvertRect es pintar un rectángulo, pero el color que toma, es el color de los píxeles pero  invertidos, que se encuentren en el HDC en cuestión. Dejando en claro que el agarra solo los píxeles del HDC en donde se va a dibujar el rectángulo con InvertRect.

¿Que significa invertir?. El color amarillo es: RGB(255,255,0) cada color es un byte. Lo que se traduce en:

Amarillo
11111111
11111111
00000000

Si invertimos
00000000
00000000
11111111

Lo que equivale en RGB(0,0,255), y aquí esta el color azul que muestra el programa, de todas manera estas operaciones booleanas lo visualizarán mas claramente con la API BitBlt.

Pie
Api: Declare Function Pie Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Los parámetros cumplen exactamente el mismo patrón que la figura mostrada en CHORD, con la diferencia que muestra lo que no muestra chord, si se dan cuenta, es como lo contrario, Chord muestra lo que Pie no muestra, y viceversa.

Polygon
Api: Declare Function Polygon Lib "gdi32" (ByVal hdc As Long, lpPoint As POINTAPI, ByVal nCount As Long) As Long

Un polígono es una figura de X cantidad de vértices y cada vértice es conectado por una recta, esto se puede ver como una definición simple de polígono, lo que hacemos en la función es pasarle como parámetros: 

lpPoint = es el arreglo que contiene a los puntos

nCount = la cantidad de puntos del polígono.

Luego Windows se encarga de conectarlos trazando una línea entre los vértices o puntos.

        coordPoly(1).X = 50

        coordPoly(1).Y = 10

        coordPoly(2).X = 25

        coordPoly(2).Y = 100

        coordPoly(3).X = 75

        coordPoly(3).Y = 100

        Polygon Me.hdc, coordPoly(1), 3

Como pueden ver en el ejemplo usamos un arreglo cuyos valores son estructuras de tipo POINT 

Si ustedes configuran un arreglo con 100 puntos y ponen 40 en el parámetro “nCount” la función solo dibujara los primeros 40 puntos del arreglo.

PolyPolygon

Api: Declare Function PolyPolygon Lib "gdi32" (ByVal hdc As Long, lpPoint As POINTAPI, lpPolyCounts As Long, ByVal nCount As Long) As Long

Se podría decir que esta función permite crear más de un polígono de manera independiente, que quiero decir, en el código lo puse como si fuera uno solo. Pero ahora pongan este código:

        coordPoly2(1).X = 50

        coordPoly2(1).Y = 10

        coordPoly2(2).X = 25

        coordPoly2(2).Y = 100

        coordPoly2(3).X = 75

        coordPoly2(3).Y = 100

        vertPoly(0) = 3

        coordPoly2(4).X = 50

        coordPoly2(4).Y = 150

        coordPoly2(5).X = 50

        coordPoly2(5).Y = 80

        coordPoly2(6).X = 100

        coordPoly2(6).Y = 120

        vertPoly(1) = 3

        PolyPolygon Me.hdc, coordPoly2(1), vertPoly(0), 2

Como segundo parámetro (lpPoint) se pasa los puntos que conforman a todos los polígonos, como tercer parámetro (lpPolyCounts) se le especifica la cantidad de puntos que posee cada polígono, y como último parámetro, se le pasa la cantidad de polígonos. Si corrieron el programa con el código cambiado habrán visto 2 triángulos. 

¿Por qué?

Configure primero 3 puntos, y vertPoly(0) = 3, con esto quiero decir que el primer polígono tiene 3 puntos. Luego configure otros 3 puntos y le dige a vertPoly(1) = 3, lo que significa que el segundo polígono también tiene 3 puntos, paso los parámetros y en el último parámetro pongo 2, y con esto le digo a la función, que todo esta adaptado para 2 polígonos.

A primera vista puede ser difícil de ver, pero hagan pruebas para que clarifiquen sus dudas.

Rectangle

Api: Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Creo que ya la hemos usado bastante para que sepan como funciona.

RoundRect

Api:  Declare Function RoundRect Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long) As Long

Función que dibuja un rectángulo cuyos bordes son redondeados; al igual que Chord usare un grafico para ilustrar los parámetros:







LÍNEAS Y CURVAS

Vamos a introducir algunas funciones que nos permitirán utilizar líneas y curvas, en este caso curvas de Bezier. Dejo claro que las siguientes funciones no funcionan en Windows 95/98/ME: ArcTo, PolyDraw y AngleArc.

Para el proyecto nada mas necesitamos un combobox, llamado Combo1, el cual colóquenlo  a la derecha del formulario y peguen este código:

Private Type POINTAPI

        X As Long

        Y As Long

End Type

Private Declare Function ArcTo Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Private Declare Function MoveToEx Lib "gdi32" (ByVal hdc As Long, ByVal X As Long, ByVal Y As Long, lpPoint As Any) As Long

Private Declare Function AngleArc Lib "gdi32" (ByVal hdc As Long, ByVal X As Long, ByVal Y As Long, ByVal dwRadius As Long, ByVal eStartAngle As Single, ByVal eSweepAngle As Single) As Long

Private Declare Function Arc Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Private Declare Function LineTo Lib "gdi32" (ByVal hdc As Long, ByVal X As Long, ByVal Y As Long) As Long

Private Declare Function PolyBezier Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, ByVal cPoints As Long) As Long

Private Declare Function PolyBezierTo Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, ByVal cCount As Long) As Long

Private Declare Function PolyDraw Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, lpbTypes As Byte, ByVal cCount As Long) As Long

Private Declare Function Polyline Lib "gdi32" (ByVal hdc As Long, lpPoint As POINTAPI, ByVal nCount As Long) As Long

Private Declare Function PolylineTo Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, ByVal cCount As Long) As Long

Private Declare Function PolyPolyline Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, lpdwPolyPoints As Long, ByVal cCount As Long) As Long

Private Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Const PT_LINETO = &H2

Const PT_BEZIERTO = &H4

Const PT_CLOSEFIGURE = &H1

Const PT_MOVETO = &H6

Dim activarinteraccion As Boolean

Dim puntosI(0 To 3) As POINTAPI

Private Sub Combo1_Click()

    Me.Cls

    Dim puntos(0 To 5) As POINTAPI

    Dim b(0 To 5) As Byte

    activarinteraccion = False

    Select Case Combo1.ListIndex

    Case 0

        MoveToEx Me.hdc, 30, 30, ByVal 0&

        LineTo Me.hdc, 150, 30

        LineTo Me.hdc, 30, 100

    Case 1

        MoveToEx Me.hdc, 100, 100, ByVal 0&

        AngleArc Me.hdc, 100, 100, 100, 0, 180

    Case 2

        Arc Me.hdc, 50, 50, 150, 150, 0, 0, 400, 50

    Case 3

        MoveToEx Me.hdc, 30, 60, ByVal 0&

        ArcTo Me.hdc, 20, 20, 50, 60, 30, 60, 20, 60

    Case 4

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = -100

        puntos(3).X = 500

        puntos(3).Y = 100

        PolyBezier Me.hdc, puntos(0), 4

    Case 5

        MoveToEx Me.hdc, 100, 25, ByVal 0&

        puntos(0).X = 125

        puntos(0).Y = 50

        puntos(1).X = 123

        puntos(1).Y = 200

        puntos(2).X = 102

        puntos(2).Y = 100

        puntos(3).X = 102

        puntos(3).Y = 100

        puntos(4).X = 312

        puntos(4).Y = 75

        puntos(5).X = 289

        puntos(5).Y = 125

        PolyBezierTo Me.hdc, puntos(0), 6

    Case 6

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = 20

        puntos(3).X = 500

        puntos(3).Y = 100

        b(0) = PT_MOVETO

        b(1) = PT_LINETO

        b(2) = PT_LINETO

        b(3) = PT_LINETO Or PT_CLOSEFIGURE

        PolyDraw Me.hdc, puntos(0), b(0), 4

    Case 7

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = 20

        puntos(3).X = 500

        puntos(3).Y = 100

        b(0) = PT_MOVETO

        b(1) = PT_BEZIERTO

        b(2) = PT_BEZIERTO

        b(3) = PT_BEZIERTO

        PolyDraw Me.hdc, puntos(0), b(0), 4

    Case 8

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = 20

        puntos(3).X = 500

        puntos(3).Y = 100

        Polyline Me.hdc, puntos(0), 4

    Case 9

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = 20

        puntos(3).X = 500

        puntos(3).Y = 100

        PolylineTo Me.hdc, puntos(0), 4

    Case 10

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = 20

        puntos(3).X = 500

        puntos(3).Y = 100

        Dim b2(0 To 4) As Long

        b2(0) = 2

        b2(1) = 2

        PolyPolyline Me.hdc, puntos(0), b2(0), 2

    Case 11

        activarinteraccion = True

        puntosI(0).X = 0: puntosI(0).Y = 0:

        puntosI(1).X = 0: puntosI(1).Y = 0:

        puntosI(2).X = 0: puntosI(2).Y = 0:

        puntosI(3).X = 0: puntosI(3).Y = 0:

    End Select

End Sub

Private Sub Form_Load()

    Combo1.AddItem "LineTo - MoveToEx"

    Combo1.AddItem "AngleArc"

    Combo1.AddItem "Arc"

    Combo1.AddItem "ArcTo"

    Combo1.AddItem "PoliBezier"

    Combo1.AddItem "PoliBezierTo"

    Combo1.AddItem "PolyDraw - PT_LINETO"

    Combo1.AddItem "PolyDraw - PT_BEZIERTO"

    Combo1.AddItem "Polylyne"

    Combo1.AddItem "Polylyneto"

    Combo1.AddItem "PolyPolyline"

    Combo1.AddItem "PoliBezier - Interactivo"

End Sub

Private Sub Form_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)

If activarinteraccion Then

    If puntosI(0).X = 0 Then

        puntosI(0).X = X \ 15: puntosI(0).Y = Y \ 15

    ElseIf puntosI(1).X = 0 Then

        puntosI(1).X = X \ 15: puntosI(1).Y = Y \ 15

    ElseIf puntosI(2).X = 0 Then

        puntosI(2).X = X \ 15: puntosI(2).Y = Y \ 15

    ElseIf puntosI(3).X = 0 Then

        Me.Cls

        puntosI(3).X = X \ 15: puntosI(3).Y = Y \ 15

        PolyBezier Me.hdc, puntosI(0), 4

        puntosI(0).X = 0: puntosI(0).Y = 0:

        puntosI(1).X = 0: puntosI(1).Y = 0:

        puntosI(2).X = 0: puntosI(2).Y = 0:

        puntosI(3).X = 0: puntosI(3).Y = 0:

    End If

End If

End Sub
Voy a explicar las funciones según el orden de colocación en el ComboBox.

MoveToEx

Api: Declare Function MoveToEx Lib "gdi32" (ByVal hdc As Long, ByVal X As Long, ByVal Y As Long, lpPoint As Any) As Long

LineTo

Api: Declare Function LineTo Lib "gdi32" (ByVal hdc As Long, ByVal X As Long, ByVal Y As Long) As Long

Estas funciones las coloco así, ya que para entenderlas hay que explicar las dos al mismo tiempo. En algunas funciones que habrán visto en el código verán que algunas terminan en TO. ¿que significa ese TO?: Windows tiene un puntero grafico,  el cual esta colocado inicialmente en el punto 0,0, ustedes dirán ¿Puntero grafico?. En una base de datos el puntero se encuentra dirigido al registro activo en ese momento. Por lo que nuestro “puntero grafico” esta dirigido al píxel en el cual Windows empezara a dibujar.

Esto tiene efecto nada mas en las funciones que terminen en TO, funciones graficas que no terminen en TO, no hacen caso a ese puntero grafico, y dibujan según las coordenadas pasadas a la función respectiva.

Ahora es mas claro ver, el por que la función LINETO solo tiene un X y un Y, ya que el otro punto para formar la recta es el “puntero grafico”, que mencione arriba.

MOVETOEX, es el API encargada de mover ese “puntero grafico”, pónganla en comentario y corran el programa verán que la línea dibujada parte desde el 0,0 hasta el primer punto 150,30. 

Esta función (MOVETOEXE) toma 4 parámetros: el dispositivo de contexto, el X y Y donde trasladamos el “puntero grafico”, y como cuarto parámetro le podemos pasar una variable de tipo POINT, el cual almacena el punto “viejo” donde estaba el “puntero grafico”

LINETO, es una función que traza una línea desde el “puntero grafico”, hasta el punto especificado en el X y Y que tiene entre sus parámetros. Como esta función tiene el TO, esto significa que esta función mueve el puntero grafico, después de dibujar la línea.

Veamos lo que hicimos:

        MoveToEx Me.hdc, 30, 30, ByVal 0& ‘Movemos el puntero grafico del 0,0 al punto 30,30 

        LineTo Me.hdc, 150, 30 ‘Dibujamos una línea desde el punto 30,30 hasta 150,30, ahora el “puntero grafico” se encuentra en 150,30.

        LineTo Me.hdc, 30, 100  ‘Dibujamos una línea desde el punto 150,30 hasta 30,100, ahora el “puntero grafico” se encuentra en 30,100.

¡claro!, espero que si ya que vamos a ver mas funciones que siguen esta metodología.

AngleArc

Api: Declare Function AngleArc Lib "gdi32" (ByVal hdc As Long, ByVal X As Long, ByVal Y As Long, ByVal dwRadius As Long, ByVal eStartAngle As Single, ByVal eSweepAngle As Single) As Long

Esta función es utilizada para hacer un circulo pero el cual tiene una parte que no se dibuja, parecido a un pacman, o una torta redonda cortada.

Entre sus parámetros tenemos:

Hdc = dispositivo de contexto en el cual se dibuja

x e y = coordenadas de donde se empieza  a dibujar la circunferencia.

dwRadius = radio de la circunferencia.

eStartAngle = Angulo en donde empieza el barrido.

eSweepAngle = Angulo que barre a partir del angulo especificado en el parámetro de arriba.

¿Qué es eso de barrer?

Viendo nuestro ejemplo puedo explicar la pregunta

        AngleArc Me.hdc, 100, 100, 100, 0, 180

Lo que estoy haciendo aquí es, un circulo de x e y igual a 100, de radio 100, y hace un barrido desde el angulo 0 hasta el angulo 180. Por esa razón solo ven la mitad de la circunferencia. Si ponen en ves de 0 colocan 90 verán ¾ partes de la circunferencia.

Arc

Api: Declare Function Arc Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Esta función como pueden ver pinta el arco de una circunferencia, los 4 primeros parámetros después del HDC determinan la ubicación de la circunferencia, los últimos cuatros determinan el arco que no se ve.

¿Cómo se determina el arco que no se ve?

Las coordenadas X3 y Y3 determinen el punto final de una recta que va desde el radio del círculo hasta ese punto. Luego el punto X4 y Y4 determinan otro punto final de otra recta que va desde el radio hasta ese punto, luego Windows de alguna manera detecta los dos puntos de intersección y no dibuja el arco que hay entre ellos.

Si están confundidos pongan este código antes de la función Arc:

MoveToEx Me.hdc, 100, 100, ByVal 0&

LineTo Me.hdc, 50, 50

MoveToEx Me.hdc, 100, 100, ByVal 0&

LineTo Me.hdc, 400, 50

Con esto deben tenerlo mas claro, la interpretación de estas cuatro líneas de arriba se las dejo a ustedes para forzarlos a entender como funciona esta función.

ArcTo

Api: Declare Function ArcTo Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Lo mismo que arc con la diferencia que mueve el “puntero grafico”.

PolyBezier
Api: Declare Function PolyBezier Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, ByVal cPoints As Long) As Long

Esta función nos grafica las curvas de Bezier, se podría decir, que esta función nos permite crear cualquier tipo de curva sin restricciones, ahora, para formar esta curva inicialmente necesitamos 4 puntos, ya que necesitamos un punto de inicio, uno final y mínimo 2 de control.

¿Qué es punto de control? 

Esos son los puntos que controlan la curva o la forma de la curva propiamente dicha, es decir, controlan al ancho de la curva, o abertura de la misma. 

Como parámetros tenemos lo siguiente: como siempre tenemos nuestro hdc, pero vean que tenemos una estructura de tipo POINTAPI, a la cual no se le pasa una estructura propiamente dicha, sino un arreglo que esta formada en su interior por esa estructura, si vemos el ejemplo:

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = -100

        puntos(3).X = 500

        puntos(3).Y = 100

Vean como cada índice posee dos variables esto se debe a que declaramos a “puntos” como un arreglo de 6 POINTAPI:

    Dim puntos(0 To 5) As POINTAPI

Ustedes dirán: ¡un momento Eduardo! este arreglo tiene 6 casillas y llenastes solo 4, pero aquí es donde juega el tercer parámetro de la función el cual es “cPoints” aquí le pasamos la cantidad de puntos que queremos que procese la función. Yo puedo tener un arreglo de 100 espacios pero puedo usar los primeros 5 y solo tengo que poner a cPoints = 5 y listo!, el agarra solo los 5 primeros espacios del arreglo y obvia al resto.

Lo que sucede es que lo declare de 6 para la función que vamos a ver mas adelantes.

Si quieren ver como puede cambiar la forma de la curva cambien los valores de los puntos 1 y 2, para que vean las diferentes formas que se crean.

NOTA: para realizar una curva de Bezier se pueden usar X cantidad de puntos es decir, puedes usar 10, 20, 30 puntos siempre que no sea 3 o 2.

PolyBezierTo
Api: Declare Function PolyBezierTo Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, ByVal cCount As Long) As Long

Exactamente la misma explicación que POLIBEZIER con la diferencia de que esta función termina en TO, lo que significa que mueve al “puntero grafico”. Y como podrán ver, aquí si uso los 6 puntos.

PolyDraw
Api: Declare Function PolyDraw Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, lpbTypes As Byte, ByVal cCount As Long) As Long

Esta es un API multifuncional, ¿por que multifuncional?, con esta API se puede usar simultáneamente LINETO, MOVETO y POLYBEZIERTO. Los parámetros son lo siguientes:

lppt = son los puntos que forman a la figura.

lpbTypes = esto se usa para darle a conocer a la función “el para que sirve el punto”.

Valores que puede tomar:

Const PT_LINETO = &H2

Const PT_BEZIERTO = &H4

Const PT_CLOSEFIGURE = &H1

Const PT_MOVETO = &H6

cCount = al igual que las anteriores funciones establece la cantidad de puntos que forman la figura.

Para entenderlo mejor veamos el ejemplo:

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = 20

        puntos(3).X = 500

        puntos(3).Y = 100

        b(0) = PT_MOVETO

        b(1) = PT_LINETO

        b(2) = PT_LINETO

        b(3) = PT_LINETO Or PT_CLOSEFIGURE

        PolyDraw Me.hdc, puntos(0), b(0), 4

¿Qué hago con el arreglo b?. Con esto le estoy diciendo. 

B(0) = PT_MOVETO significa que la acción es “mueve el puntero grafico al punto 10,100”, noten que B(0) se corresponde al índice puntos(0).

B(1) = PT_LINETO con esto le digo traza una recta desde el “puntero grafico” hasta 200,300”

Y así con B(2) y B(3) pero en B(3) además le digo que me cierre la figura. Vean que aparece la palabra TO, por lo que deben de entender que el “puntero grafico” se esta moviendo.

Fíjense que esta función traza un polígono con una solo llamada , si lo hiciéramos por separado tendríamos que llamar 1 vez al API MOVETOEX y 3 veces LINETO.

Y en el “case 7” tenemos la misma función pero con Bezier:

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = 20

        puntos(3).X = 500

        puntos(3).Y = 100

        b(0) = PT_MOVETO

        b(1) = PT_BEZIERTO

        b(2) = PT_BEZIERTO

        b(3) = PT_BEZIERTO

        PolyDraw Me.hdc, puntos(0), b(0), 4

Aquí configuramos 3 puntos como PT_BEZIER, ustedes podrían preguntar ¿Dónde esta el punto de inicio? El esta donde esta inicialmente el “puntero Grafico” que en este caso es 10, 100 ya que la primera acción b(0) = PT_MOVETO.

Es decir, con el primer arreglo, ustedes configuran los puntos, y con el segundo configuran para que sirve cada punto. Y luego la función hace el resto. Lo único malo es que estas función no esta habilitada para Windows 98/95/ME, sino para 2000/NT/XP.

Polyline
Api: Declare Function Polyline Lib "gdi32" (ByVal hdc As Long, lpPoint As POINTAPI, ByVal nCount As Long) As Long

Los parámetros cumplen la misma función que en POLIBEZIER con la diferencia que ahora en ves de formar una curva, trazan una recta que empieza desde “puntos(0)”, va pasando por los demas puntos y termina en “puntos(3)”. Creo que es fácil de entender si entendieron Bezier.

PolylineTo
Api: Private Declare Function PolylineTo Lib "gdi32" (ByVal hdc As Long, lpPoint As POINTAPI, ByVal nCount As Long) As Long

Igual que POLYLINE con la diferencia que mueve el “puntero grafico”.

PolyPolyline
Api: Private Declare Function PolyPolyline Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, lpdwPolyPoints As Long, ByVal cCount As Long) As Long

Esta función al igual con POLYLINE sirve para crear polígonos, la diferencia es que puedes crear más de un polígono. ¿Qué quiero decir? Se recuerdan de la función PolyDraw, bueno es más o menos el mismo sentido. Para entenderlo mejor expliquemos con el ejemplo:

        puntos(0).X = 10

        puntos(0).Y = 100

        puntos(1).X = 200

        puntos(1).Y = 300

        puntos(2).X = 300

        puntos(2).Y = 20

        puntos(3).X = 500

        puntos(3).Y = 100

        Dim b2(0 To 4) As Long

        b2(0) = 2

        b2(1) = 2

        PolyPolyline Me.hdc, puntos(0), b2(0), 2

Cuando ustedes arrancan el programa ¿Qué ven? Dos líneas, como dije arriba esta función puede pintar polígonos independientes.

Veamos los parámetros:

lppt = son los puntos que forman a la figura.

lpdwPolyPoints = este es un arreglo que determina cuantas figuras vamos a manejar y la cantidad de puntos que manejan.

nCount = establece la cantidad de figuras que graficara la función.

¿Por qué vemos 2 líneas? 

Observen que en el arreglo “puntos” configure 4 puntos, y en el arreglo “b2” solo configure 2 casillas o espacios de 4 que originalmente había inicializado en DIM, con esto le digo a la función que mi matriz “puntos” consta de 4 coordenadas (x,y) para 2 figuras.

Cuando hago b2(0) = 2, le estoy diciendo a la función que la primera figura esta formada por dos puntos, y b2(1) = 2, le estoy diciendo lo mismo.

Luego en nCount = 2, y aquí le digo que tome del arreglo solo dos casilla o espacios.

Se que puede sonar, engorroso, complicado, igual fue para mi entender esto por primera vez, pero inventen cositas y verán que entenderán  todas estas funciones rápidamente.

NOTA: la opción de "PoliBezier - Interactivo", para que funcione tiene que darle clic 4 veces a la pantalla en puntos distintos, cuando le den por cuarta vez, el programa le pintara una curva de bezier, formada por esos 4 puntos. Fue algo adicional que quise colocar.

CAMINOS (PATH)
Imaginemos que tenemos una imagen en un HDC, esa imagen la queremos recortar, si fuera un recorte rectangular sería fácil con el API BitBlt que veremos mas adelante se puede hacer, pero imaginen que quieren recortar la imagen pero con formas raras, como curvas, elipses, círculos, o polígonos irregulares, sería algo grandioso, pero programar eso sería muy complicado y haría de nuestro código algo engorroso y feo, pero Windows tiene una solución para ello y es “Caminos”, “Caminos” no es mas que trazar mediante una herramienta de dibujo un camino para luego manipularlo, como por ejemplo recortar el camino y pintar el área enmarcada.

Las funciones API’S son pocas y sencillas así que vamos al ejemplo, este programa sigue la filosofía del programa de “Sistema de coordenadas”, me refiero a la captura del mensaje WM_PAINT, pero para que funcione pongan en la carpeta del proyecto la imagen “image.bmp”, que acompaña a este manual.

Luego en el formulario pongan este código:

Private Sub Form_Click()

Unload Me

End Sub

Private Sub Form_Load()

    Cambiar_Proc Me.hwnd

End Sub

En el modulo este código:

Option Explicit

Const LR_LOADFROMFILE = &H10

Const IMAGE_BITMAP = 0

Public Type POINTAPI

        x As Long

        y As Long

End Type

Public Type RECT

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

Public Type PAINTSTRUCT

        hdc As Long

        fErase As Long

        rcPaint As RECT

        fRestore As Long

        fIncUpdate As Long

        rgbReserved(32) As Byte

End Type

Public Const RGN_AND = 1

Public Const RGN_COPY = 5

Public Const RGN_DIFF = 4

Public Const RGN_OR = 2

Public Const RGN_XOR = 3

Declare Function BeginPath Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function EndPath Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function StrokePath Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function FillPath Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function StrokeAndFillPath Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function SelectClipPath Lib "gdi32" (ByVal hdc As Long, ByVal iMode As Long) As Long

Declare Function CloseFigure Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function WidenPath Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function CreateCompatibleDC Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal dwImageType As Long, ByVal dwDesiredWidth As Long, ByVal dwDesiredHeight As Long, ByVal dwFlags As Long) As Long

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Declare Function BeginPaint Lib "user32" (ByVal hwnd As Long, lpPaint As PAINTSTRUCT) As Long

Declare Function EndPaint Lib "user32" (ByVal hwnd As Long, lpPaint As PAINTSTRUCT) As Long

Declare Function BitBlt Lib "gdi32" (ByVal hDestDC As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal dwRop As Long) As Long

Declare Function DeleteDC Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function PolyBezier Lib "gdi32" (ByVal hdc As Long, lppt As POINTAPI, ByVal cPoints As Long) As Long

Declare Function Polygon Lib "gdi32" (ByVal hdc As Long, lpPoint As POINTAPI, ByVal nCount As Long) As Long

Public Const WM_MOUSEMOVE = &H200

Private Const WM_PAINT = &HF

Public Const GWL_WNDPROC = -4

Public nuevoproc As Long

Public viejoproc As Long

Public mensageForm As String

Public brocha As Long

Public hdc As Long

Dim bitmap As Long

Dim dc As Long

Dim puntos(0 To 3) As POINTAPI

Sub Cambiar_Proc(hwnd2 As Long)

nuevoproc = ObtenerLong(AddressOf MenuProc)

viejoproc = SetWindowLong(hwnd2, GWL_WNDPROC, nuevoproc)

End Sub

Function ObtenerLong(ByVal direcfunt As Long) As Long

    ObtenerLong = direcfunt

End Function

Private Function MenuProc(ByVal hwnd As Long, ByVal message As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    Dim coordPoly(1 To 3) As POINTAPI

    If message = WM_PAINT Then

        Dim p As PAINTSTRUCT

        hdc = BeginPaint(hwnd, p)

        brocha = CreateSolidBrush(RGB(200, 200, 200))

        SelectObject hdc, brocha

        Rectangle hdc, 0, 0, form1.Width, form1.Height

        DeleteObject brocha

        BeginPath hdc

        'coordPoly(1).x = 50

        'coordPoly(1).y = 10

        'coordPoly(2).x = 25

        'coordPoly(2).y = 100

        'coordPoly(3).x = 200

        'coordPoly(3).y = 400

        'Polygon hdc, coordPoly(1), 3

        puntos(0).x = 10

        puntos(0).y = 100

        puntos(1).x = 200

        puntos(1).y = 300

        puntos(2).x = 300

        puntos(2).y = -100

        puntos(3).x = 500

        puntos(3).y = 100

        PolyBezier hdc, puntos(0), 4

        EndPath hdc

        CloseFigure hdc

        SelectClipPath hdc, RGN_COPY

        'brocha = CreateSolidBrush(RGB(200, 200, 0))

        'SelectObject hdc, brocha

        'FillPath hdc

        'StrokePath hdc

        'StrokeAndFillPath hdc

        'DeleteObject brocha

        bitmap = LoadImage(App.hInstance, App.Path & "\image.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

        dc = CreateCompatibleDC(hdc)

        SelectObject dc, bitmap

        BitBlt hdc, 0, 0, 800, 600, dc, 0, 0, vbSrcCopy

        DeleteDC dc

        EndPaint hwnd, p

    Else

        MenuProc = CallWindowProc(viejoproc, hwnd, message, wParam, lParam)

    End If

End Function

Algunas líneas están puesta en comentario a propósito, ya que en el transcurso de la explicación iremos quitando esos comentarios y poniendo otros para explicar las funciones. En este código vuelven aparecer funciones API’S nuevas que no entran por completo en el tema,  estas son : LOADIMAGE, CREATECOMPATIBLEDC, BITBLT, DELETEDC, la única que no voy a explicar ya que tiene una sección especial es BITBLT, el resto debido a que no tengo una sección especifica para ellas las voy explicar a continuación.

Antes de entrar a la explicación de “Caminos”,  voy explicar las funciones LOADIMAGE, CREATECOMPATIBLEDC y DELETEDC.

Antes de empezar ustedes se podrían preguntar ¿Por qué no cargastes la imagen en la propiedad Picture del formulario?. Bueno tan simple como recordarles que nosotros en este código nos encargamos de pintar nuestro formulario al 100%, ¿qué quiero decir?. Que no le dejamos a la función de mensajes de VB procesar el mensaje WM_PAINT, si ustedes colocan la imagen por la propiedad picture del formulario se darán cuenta que la imagen no aparece, lo que es lógico ya que el mensaje WM_PAINT lo capturamos nosotros y no la función de  mensajes que pone VB. ¿como hacemos entonces?. Aquí es donde entran las 4 funciones arriba mencionadas.

LoadImage

Api: Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal dwImageType As Long, ByVal dwDesiredWidth As Long, ByVal dwDesiredHeight As Long, ByVal dwFlags As Long) As Long

Creo que el nombre de la función lo dice todo, carga una imagen en memoria, fíjense que no tiene un HDC ya que la función retorna la referencia a la imagen cargada. Que luego ubicaremos en un HDC pero eso ya lo veremos.

Parámetros:

hInst = aquí especificamos el handle de la instancia de nuestra aplicación.

lpsz =  en caso de que la imagen este en un archivo de recursos, pues aquí se le coloca el nombre o ID del recurso especifico, pero este no es nuestro caso, ya que nosotros estamos cargando la imagen a través de un fichero, así que, para nosotros este parámetro toma la dirección de la imagen. Lo del recurso lo menciono para aquellos que usen C++.

dwImageType: especifica el tipo de imagen a cargar:

Const IMAGE_BITMAP = 0

Const IMAGE_ICON = 1

Const IMAGE_CURSOR = 2

dwDesiredWidth  y dwDesiredHeight = especifica el ancho y alto de la imagen que se va a cargar.

dwFlags = puede tomar los valores que se muestran a continuación:

LR_DEFAULTCOLOR = valor por defecto y no tiene ningún significado.

LR_CREATEDIBSECTION  = solo aplicable para imágenes bitmap, y hace que la función retorne una imagen independiente de dispositivo.

LR_DEFAULTSIZE = Se utiliza para decirle al sistema las dimensiones que debe de tomar la imagen.

LR_LOADFROMFILE = Utilizado cuando se quiere recuperar una imagen ubicada o contenida en un archivo.

LR_MONOCHROME = carga la imagen en blanco y negro.

Hay mas pero considero que estos son los mas importantes, en la documentación oficial encontrarán los restantes.

Si verificamos en el código:

        bitmap = LoadImage(App.hInstance, App.Path & "\image.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

Podemos ver que estamos cargando una imagen desde un fichero, por el valor LR_LOADFROMFILE, a las dimensiones les paso 0 para que tome las dimensiones completa de la imagen.

Creo que es sencillo de ver el funcionamiento de esta función. Pero vean que “bitmap” toma el valor de referencia de la imagen ¿Cómo hacemos para pegarla en nuestro HDC?. Con la función que explico a continuación solucionamos el problema.

CreateCompatibleDC

Api: Declare Function CreateCompatibleDC Lib "gdi32" (ByVal hdc As Long) As Long

Nosotros no podemos cargar una imagen y pegarla directamente a nuestro HDC del formulario, hay que usar un HDC intermedio donde almacenemos la imagen para luego pasarla a nuestro HDC original. Esta función lo que hace es crear un DC compatible de uno ya existente.

¿Qué significa compatible?. Se recuerdan que yo les dije que un HDC o dispositivo de contexto viene con objetos configurados (pincel, brocha, fuentes, etc), cuando crean un DC o HDC con esta función lo que están haciendo es creando un HDC que tenga TODOS los objetos tal cual como el HDC original, es esa la razón de que solo la función tome un parámetro y es el HDC original del cual queremos tomar esos objetos.

        dc = CreateCompatibleDC(hdc)

En el código vemos esto, aquí la variable “dc” esta tomando los objetos de la variable “hdc” que es nuestro dispositivo de contexto original, es decir, de nuestro formulario. Ahora la variable “dc” es un dispositivo de contexto, ya que tomo de “hdc”, la brocha, pinceles, fuentes, etc, y todo lo que tenga nuestro “hdc” original.

Ahora un bitmap es un objeto, si queremos que el HDC creado (dc) tome la imagen pues usemos la siguiente línea:

        SelectObject dc, bitmap

Aquí estamos pegando el objeto “bitmap” a nuestra hoja de dibujo (HDC). Pero aquí no acaba todo, ya que nosotros queremos que la imagen este en la variable “hdc” que es el dispositivo de contexto del formulario, y para ello usamos una función para copiar imágenes:

        BitBlt hdc, 0, 0, 800, 600, dc, 0, 0, vbSrcCopy

Esta función tiene un apartado en la sección de “Imágenes” que se ve mas adelante.

DeleteDC

Api: Declare Function DeleteDC Lib "gdi32" (ByVal hdc As Long) As Long

Un DC o HDC (OJO: en realidad el dispositivo de contexto su abreviatura es DC no HDC, el problema es que uso HDC ya que ese el nombre que le da VB en sus objetos, pero me refiero a lo mismo cuando hablo de DC o HDC), es un objeto particular que tiene su propia función para eliminarlos, los DC que creamos hay que eliminarlos, sino quedan en memoria y VB no los borra, para eliminarlo usamos esta función, que se usa solo para DC.

        DeleteDC dc

En nuestro código uso esta función para eliminar la variable “dc” después de haberla utilizado.

Ya explicado estas 3 funciones que permiten la inserción de una imagen ya podemos entrar a las funciones de este tema. Vean que no es tan fácil insertar una imagen, como nosotros creemos, lo que sucede es que los lenguajes en general esconde esta parte del código, por ejemplo en VB es mas fácil insertar una imagen ya que le asigna a la propiedad “Picture” la imagen y listo!. Pero ya se pueden imaginar mas o menos lo que sucede en su interior.

BeginPath

Api: Declare Function BeginPath Lib "gdi32" (ByVal hdc As Long) As Long

EndPath

Api: Declare Function EndPath Lib "gdi32" (ByVal hdc As Long) As Long

Estas funciones son las encargadas de empezar y finalizar un camino, cuando queremos trazar un camino tenemos que usar BeginPath pasándole como parámetro el hdc donde se trazara, luego se emplean las funciones de dibujo para trazar el camino, y cuando se haya terminado, este se cierra con EndPath pasándole el hdc como parámetro y así le indicamos a Windows que el camino a terminado.

Las funciones para crear caminos se usan dependiendo del sistema operativo, a continuación se muestra las funciones disponibles dependiendo de la versión de Windows.

Windows 2000/NT:

AngleArc, Arc, ArcTo, Chord, CloseFigure, Ellipse, ExtTextOut, LineTo, MoveToEx, Pie, PolyBezier, PolyBezierTo, PolyDraw, Polygon, Polyline, PolyLineTo, PolyPolygon, PolyPolyline, Rectangle, RoundRect, TextOut.

Windows 95/98/ME:

CloseFigure, ExtTextOut, LineTo, MoveToEx, PolyBezier, PolyBezierTo, Polygon, Polyline, PolylineTo, PolyPolygon, PolyPolyLine, TextOut.

Es por esa razón que en entre BeginPath y EndPath tengo la función Polygon y PoliBezier. Ya que quería hacerlo compatible con cualquier versión del sistema operativo.

¿Qué hacemos con el camino?. Ya tenemos el camino trazado, ahora vamos a ver las funciones que me permiten manipular los caminos, si corrieron el programa vieron que se dibuja la curva de Bezier cuyo fondo es parte de la imagen. Para ese efecto usamos la función siguiente.

SelectClipPath

Api: Declare Function SelectClipPath Lib "gdi32" (ByVal hdc As Long, ByVal iMode As Long) As Long

Esta función aplica una operación booleana, entre el HDC que se pasa como parámetro y el camino antes trazado, esa operación booleana se especifica en el parámetro “iMode”, que puede tomar las valores que se muestran a continuación:

Public Const RGN_AND = 1 (INTERSECCION)

Public Const RGN_COPY = 5 (COPIA)

Public Const RGN_DIFF = 4 (DIFERENCIA)

Public Const RGN_OR = 2 (UNION)

Public Const RGN_XOR = 3 (UNION EXCEPTO DE LAS AREAS COMUNES, QUE SE EXCLUYEN)

En nuestra caso usamos RGN_COPY, otro interesante es RGN_DIFF, pónganlo para que vean el efecto, que es como “inverso” a RGN_COPY.

CloseFigure

Api: Declare Function CloseFigure Lib "gdi32" (ByVal hdc As Long) As Long

Esta función como su nombre lo indica “cierra una figura”, vean que la curva PoliBezier que use, esta cerrada por una línea recta entre el punto inicial y el punto final, cuando un polígono cualquiera se dibuja y no es cerrado, al usar esta función la figura es cerrada trazando una recta desde el punto inicial al punto final. Como único parámetro se le pasa el HDC en donde se esta dibujando.

Ahora en el código vamos hacer esto, pongan en comentario lo que esta en azul y quítenle el comentario a lo que esta en rojo:

         ‘SelectClipPath hdc, RGN_DIFF

        brocha = CreateSolidBrush(RGB(200, 200, 0))

        SelectObject hdc, brocha

        FillPath hdc

        ‘StrokePath hdc

        ‘StrokeAndFillPath hdc

        DeleteObject brocha

        ‘bitmap = LoadImage(App.hInstance, App.Path & "\image.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

       ‘dc = CreateCompatibleDC(hdc)

        ‘SelectObject dc, bitmap

        ‘BitBlt hdc, 0, 0, 800, 600, dc, 0, 0, vbSrcCopy

        ‘DeleteDC dc

Cuando corran el programa verán que muestra el área de la curva rellena con un color. Esto es gracia a la siguiente función.

FillPath

Api: Declare Function FillPath Lib "gdi32" (ByVal hdc As Long) As Long

Esta función rellena los caminos trazados con la brocha actual, en mi caso yo hice una nueva brocha para que vieran el efecto. Como parámetro se le pasa el hdc en donde estamos dibujando.

Ahora pongan en comentario la línea donde esta FillPath, y quítenle el comentario a “StrokePath”.

StrokePath

Api: Declare Function StrokePath Lib "gdi32" (ByVal hdc As Long) As Long

Esta función solo traza la figura que dibujamos ¡no le rellena!, ni siquiera la cierra como pueden ver. Al igual que la anterior su único parámetro es el hdc.

Ahora pongan en comentario la línea donde esta StrokePath, y quítenle el comentario a “StrokeAndFillPath”.

StrokeAndFillPath

Api: Declare Function StrokeAndFillPath Lib "gdi32" (ByVal hdc As Long) As Long

Esta  función es una mezcla de las 2 anteriores, es decir, rellena y traza la figura al mismo tiempo, en este caso cerrándola si esta abierta. Su único parámetro es un hdc.

WidenPath

Api: Declare Function WidenPath Lib "gdi32" (ByVal hdc As Long) As Long

Esta API que no uso en el código hace lo siguiente. Recuerdan que ustedes para trazar la figura pueden usar un pincel con un grueso mayor que 1, eso lo vimos en la parte de pinceles, esta función (WidenPath) se utiliza cuando se trazan caminos con pinceles de grueso mayor que 1, sirve para 1, pero no se ve el efecto ya que la línea es muy delgada.

¿Qué hace esta función?. Para que funcione WidenPath tienen que mostrar la imagen de nuevo, con la diferencia que la porción de la imagen que se muestra no es el área de la figura, sino el grueso del pincel. ¿Están confundido, verdad?. Vean el cuadro.


Con la función SelectClipPath la parte que esta en azul se rellanaba con la imagen, esto ya lo probamos, ahora WindPath lo que hace es rellenar la parte negra con la imagen que ahí se encuentre y la parte azul no lo hace nada ¿Ahora lo ven?. Esa es la razón del por que tiene que usar un pincel con un grosor considerable.

REGIONES

A continuación, se presenta un tema muy bonito, en lo que a librerías graficas se refiere, ya que las funciones que veremos a continuación presentan un potencial muy grande para desarrollar nuestras aplicaciones.

Las regiones son parecidas a los caminos, pero mucho mas complejas y completas, caminos nos permitía crear secciones o áreas, dependiendo de una figura ya estipulada, pero nada mas!, en cambio regiones se trabaja solo con áreas, y no solo áreas simples como trazábamos en caminos sino áreas que pueden ser muy complejas y también nos permite la manipulación de  pulsaciones del mouse. Es decir, imaginen un triangulo, ustedes quieren que haga algo cuando se pulse el botón del mouse sobre él, hacerlo con las herramientas de VB es complicado ya que hay que ver las coordenadas de pulsación compararlas con los vértices, etc, etc, etc..., con regiones este trabajo es fácil. Imaginen crear un mapa de países, donde cada país es una forma irregular, con las herramientas conocidas sería un arduo trabajo programar todo eso, incluyendo las pulsaciones del mouse.

Las funciones referentes a regiones son bastante por lo que se va dividir este tema en varios proyectos, en uno de ellos donde explico lo básico y el otro donde trabajo con imágenes; es con regiones donde veremos formularios irregulares,  verán como hacemos un formulario en forma irregular y luego usaremos una imagen para crear nuestros formularios.

En esta sección voy a empezar al revés, es decir, primero voy a explicar algunas funciones API’S referente a regiones antes de colocar el código. Las regiones no van ligadas directamente al HDC o al HWDN, es decir, las funciones para crear las regiones no tienen como parámetro un HDC, ya que primero se crea una región y luego se le asigna a un HDC o HWND.

Para crear las regiones, no es como los caminos, que usaban las funciones graficas de Windows, para crearlos, las regiones tienen sus propias funciones para trazar las figuras.

CreateRectRgn

Api: Declare Function CreateRectRgn Lib "gdi32" (ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

CreateRectRgnIndirect

Api: Declare Function CreateRectRgnIndirect Lib "gdi32" (lpRect As RECT) As Long

CreateRoundRectRgn

Api:Declare Function CreateRoundRectRgn Lib "gdi32" (ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long) As Long

Estas 3 funciones sirven para crear regiones cuya forma es rectangular, las dos primeras hacen exactamente lo mismo, con la diferencia que una usa la estructura RECT y la otra usa las 4 coordenadas necesarias para crear un rectángulo. La  tercera función tiene el mismo efecto que la función RoundRect.

CreateEllipticRgn

Api: Declare Function CreateEllipticRgn Lib "gdi32" (ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

CreateEllipticRgnIndirect

Api: Declare Function CreateEllipticRgnIndirect Lib "gdi32" (lpRect As RECT) As Long

CreatePolygonRgn

Declare Function CreatePolygonRgn Lib "gdi32" (lpPoint As POINTAPI, ByVal nCount As Long, ByVal nPolyFillMode As Long) As Long

CreatePolyPolygonRgn

Declare Function CreatePolyPolygonRgn Lib "gdi32" (lpPoint As POINTAPI, lpPolyCounts As Long, ByVal nCount As Long, ByVal nPolyFillMode As Long) As Long

Las primeras dos funciones, es fácil ver lo que hacen, crean una región en forma de elipse, y las ultimas 2 crean polígonos.

Vean que los parámetros son exactamente los mismos que sus homólogos presentados en la sección de “Figuras Geométricas” y “Líneas y Curvas”. A diferencia de caminos que teníamos todas las funciones graficas a nuestra disposición, con regiones solo podemos crear regiones de las formas arriba mencionadas. Dense cuenta que entre sus parámetro no existe un HDC.

Proyecto 1

Ya mostrada estas funciones básicas, ya podemos ir al ejemplo, que consta de un formulario. En él introducen este código:

Private Type POINTAPI

        X As Long

        Y As Long

End Type

Private Type RECT

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

 Const RGN_AND = 1

 Const RGN_COPY = 5

 Const RGN_DIFF = 4

 Const RGN_OR = 2

 Const RGN_XOR = 3

Private Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Private Declare Function CreateRectRgn Lib "gdi32" (ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function CreatePolygonRgn Lib "gdi32" (lpPoint As POINTAPI, ByVal nCount As Long, ByVal nPolyFillMode As Long) As Long

Private Declare Function CombineRgn Lib "gdi32" (ByVal hDestRgn As Long, ByVal hSrcRgn1 As Long, ByVal hSrcRgn2 As Long, ByVal nCombineMode As Long) As Long

Private Declare Function FillRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long, ByVal hBrush As Long) As Long

Private Declare Function FrameRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long, ByVal hBrush As Long, ByVal nWidth As Long, ByVal nHeight As Long) As Long

Private Declare Function InvertRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long) As Long

Private Declare Function PtInRegion Lib "gdi32" (ByVal hRgn As Long, ByVal X As Long, ByVal Y As Long) As Long

Dim coordPoly(1 To 3) As POINTAPI

Dim region(0 To 4) As Long

Private Sub Form_DblClick()

    Unload Me

End Sub

Private Sub Form_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)

    Dim i As Integer

    Dim resultado As Long

    For i = 0 To UBound(region)

        resultado = PtInRegion(region(i), X / 15, Y / 15)

        If resultado <> 0 Then

            MsgBox "Se le ha dada click a la region " & i, vbInformation + vbOKOnly, ""

            Exit For

        End If

    Next

End Sub

Private Sub Form_Paint()

        region(0) = CreateRectRgn(10, 10, 100, 100)

        coordPoly(1).X = 200

        coordPoly(1).Y = 10

        coordPoly(2).X = 255

        coordPoly(2).Y = 100

        coordPoly(3).X = 150

        coordPoly(3).Y = 100

        region(1) = CreatePolygonRgn(coordPoly(1), 3, WINDING)

        coordPoly(1).X = 200

        coordPoly(1).Y = 200

        coordPoly(2).X = 255

        coordPoly(2).Y = 150

        coordPoly(3).X = 150

        coordPoly(3).Y = 150

        region(2) = CreatePolygonRgn(coordPoly(1), 3, WINDING)

        CombineRgn region(0), region(0), region(1), RGN_OR

        CombineRgn region(0), region(0), region(2), RGN_OR

        FillRgn Me.hdc, region(0), CreateSolidBrush(RGB(0, 255, 200))

        'FillRgn Me.hdc, region(0), CreateSolidBrush(RGB(0, 255, 200))

        'FillRgn Me.hdc, region(1), CreateSolidBrush(RGB(0, 255, 200))

        'FillRgn Me.hdc, region(2), CreateSolidBrush(RGB(0, 255, 200))

        'FrameRgn Me.hdc, region(0), CreateSolidBrush(RGB(0, 255, 100)), 10, 10

        'InvertRgn Me.hdc, region(0)

End Sub

CombineRgn
Api: Declare Function CombineRgn Lib "gdi32" (ByVal hDestRgn As Long, ByVal hSrcRgn1 As Long, ByVal hSrcRgn2 As Long, ByVal nCombineMode As Long) As Long

Esta función es utilizada para combinar regiones. Como primer parámetro “hDestRgn”, pasamos la variable donde almacenaremos la nueva región, creado a partir de los dos siguientes parámetros que también son regiones, “hSrcRgn1” y “hSrcRgn2”, y como último parámetro “nCombineMode”, le pasamos que tipo de operación lógica va a realizar, los valores que puede tomar son los siguientes:

 Const RGN_AND = 1

 Const RGN_COPY = 5

 Const RGN_DIFF = 4

 Const RGN_OR = 2

 Const RGN_XOR = 3

Que si se dan cuenta son los mismos que en “caminos”.

Veamos el ejemplo:

       CombineRgn region(0), region(0), region(1), RGN_OR

       CombineRgn region(0), region(0), region(2), RGN_OR

Estas son las líneas usadas en el código, region(0) corresponde inicialmente a un rectángulo, y region(1) y region(2) son 2 triángulos con distintas coordenadas. Noten que estoy almacenando las combinaciones en region(0). Y uso el operador lógico RGN_OR.

En la primera linea almaceno en region(0), la combinación de la region(0) inicial y el triangulo en la region(1), y luego combino lo que tiene ahora region(0) con el otro triangulo trazado en region(2). Y como resultado tengo 3 regiones almacenadas, por decirlo así, en una sola variable.

Esta función puede retornar valores, que pueden ser lo siguientes:

Public Const NULLREGION = 1 ‘ La región resultante es nula

Public Const SIMPLEREGION = 2 ‘ La región resultante es simple

Public Const COMPLEXREGION = 3 ‘ La región resultante es una compleja

Esto no lo he probado pero he de suponer que Windows llama a SIMPLEREGION regiones como rectángulos, elipse o polígonos simples, y habla de complejas a regiones que pueden ser combinación de ambas. Les dejo a ustedes probar eso.

¿Qué podemos hacer con las regiones? Observen  las siguientes funciones, vean que todavía no hemos usado ningún dispositivo de contexto (HDC) o ventana (HWND).

FillRgn

Api: Declare Function FillRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long, ByVal hBrush As Long) As Long

Esta función rellena las regiones con una brocha especificada en el parámetro “hBrush”, noten que aquí si hay un “hdc”, que es el dispositivo de contexto que contiene la region o regiones, como segundo parámetro se le pasa la región que se quiere rellenar.

En el código:

FillRgn Me.hdc, region(0), CreateSolidBrush(RGB(0, 255, 200))

Vean que creo la brocha de manera directa, es decir, pongo la función CreateSolidBrush directo, evitándome así declarar variables que en algunos casos son necesarias pero otra veces no. Como segundo parámetro paso la region(0), que es la region a rellenar.

La función que muestro a continuación no esta en el código, pero hace exactamente lo mismo que FillRgn, con la diferencia que la brocha que toma es la que tiene configurado el dispositivo de contexto (HDC).

PaintRgn

Api: Declare Function PaintRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long) As Long

Como dije arriba hace lo mismo pero la brocha que usa es la del hdc, que se le pasa como primer parámetro.

FrameRgn

Api: Declare Function FrameRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long, ByVal hBrush As Long, ByVal nWidth As Long, ByVal nHeight As Long) As Long

Para ver el efecto de esta función, tienen que quitarle el comentario a esta línea en el código:

FrameRgn Me.hdc, region(0), CreateSolidBrush(RGB(0, 255, 100)), 10, 10

Y ponerle comentario a la linea:

'FillRgn Me.hdc, region(0), CreateSolidBrush(RGB(0, 255, 200))

Que era la linea que estaba activada antes. Se darán cuenta rápidamente de lo que hace la función cuando arrancan el programa, sus 3 primeros parámetros son idénticos que FillRgn, con la diferencia que la brocha no rellena la figura sino los bordes de la figura, cuyo grosor es controlado por las 2 ultimas variables “nWidth” que controla el grosor de ancho y “nHeight” controla el grosor de alto.

InvertRgn

Api: Declare Function InvertRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long) As Long

Para ver el efecto de la función al igual que arriba ponen en comentario esta linea

‘FrameRgn Me.hdc, region(0), CreateSolidBrush(RGB(0, 255, 100)), 10, 10

Y le quitan el comentario a:

InvertRgn Me.hdc, region(0)

Esta función es sencilla lo que hace es invertir los colores, es decir, si el HDC pasado en el primer parámetro tiene un color blanco, pues la region será negra, y si el fondo es negro pues la region será blanca.

PtInRegion

Api: Declare Function PtInRegion Lib "gdi32" (ByVal hRgn As Long, ByVal X As Long, ByVal Y As Long) As Long

Aquí presentamos la función que le da una versatilidad y poder al uso de regiones,  que es la posibilidad de saber si ha dado clic a una región, corran el programa y denle clic a cualquier región, notaran que le aparece un mensaje indicándole la región a la cual se ha dado clic, siempre  aparece el número 0 y eso por que recuerden que combinamos las regiones. Pongan en comentario estas líneas:

        ‘CombineRgn region(0), region(0), region(1), RGN_OR

        ‘CombineRgn region(0), region(0), region(2), RGN_OR

Y quítenle el comentario a estas líneas:

        FillRgn Me.hdc, region(0), CreateSolidBrush(RGB(0, 255, 200))

        FillRgn Me.hdc, region(1), CreateSolidBrush(RGB(0, 255, 200))

        FillRgn Me.hdc, region(2), CreateSolidBrush(RGB(0, 255, 200))

Lo que estoy haciendo es pintar las regiones una a una sin combinarlas
. Corran el programa y denle clic a las regiones, verán que el mensaje indica el número de region en donde fue pulsado el ratón, ¡no les parece asombroso!, la posibilidad de crear regiones complejas y que nos brinden la posibilidad de capturar el clic del mouse. Imaginen la creación de un mapa, donde podamos capturar los clic de los mapas con una precisión exacta.

Esta función se encuentra en el código así:

    Dim i As Integer

    Dim resultado As Long

    For i = 0 To UBound(region)

        resultado = PtInRegion(region(i), X / 15, Y / 15)

        If resultado <> 0 Then

            MsgBox "Se le ha dada click a la region " & i, vbInformation + vbOKOnly, ""

            Exit For

        End If

    Next

Lo que hago aquí es verificar en mi arreglo de regiones, cual region fue seleccionado por el clic del mouse. La función PtInRegion funciona de la siguiente manera: como primer parámetro le pasamos la región a evaluar, como segundo y tercer parámetro le pasamos las coordenadas en donde se dio clic. Estas coordenadas son en PIXELES es por esa razón que divido entre 15.

Si la coordenada X e Y se encuentran un una región, la variable “resultado” tiene un valor distinto de 0, en caso de que una región no se encuentre en las coordenadas pasadas, retorna 0.

Proyecto 2

Vamos a mostrar el segundo proyecto sobre regiones, que trata sobre dos funciones, que bien empleadas pueden hacer cosas curiosas. Para este proyecto es necesario un formulario y que la imagen “image.bmp” se encuentre en la carpeta en donde esta el proyecto.

Luego pegar este código:

Const LR_LOADFROMFILE = &H10

Const IMAGE_BITMAP = 0

Private Type POINTAPI

        X As Long

        Y As Long

End Type

Private Type RECT

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

 Const RGN_AND = 1

 Const RGN_COPY = 5

 Const RGN_DIFF = 4

 Const RGN_OR = 2

 Const RGN_XOR = 3

Private Declare Function CreateEllipticRgn Lib "gdi32" (ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function OffsetRgn Lib "gdi32" (ByVal hRgn As Long, ByVal X As Long, ByVal Y As Long) As Long

Private Declare Function GetClientRect Lib "user32" (ByVal hwnd As Long, lpRect As RECT) As Long

Private Declare Function StretchBlt Lib "gdi32" (ByVal hdc As Long, ByVal X As Long, ByVal Y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal nSrcWidth As Long, ByVal nSrcHeight As Long, ByVal dwRop As Long) As Long

Private Declare Function SelectClipRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long) As Long

Private Declare Function ExtSelectClipRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long, ByVal fnMode As Long) As Long

Private Declare Function CreateCompatibleDC Lib "gdi32" (ByVal hdc As Long) As Long

Private Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal dwImageType As Long, ByVal dwDesiredWidth As Long, ByVal dwDesiredHeight As Long, ByVal dwFlags As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteDC Lib "gdi32" (ByVal hdc As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Dim coordPoly(1 To 3) As POINTAPI

Dim region(0 To 4) As Long

Dim bitmap As Long

Dim cx As Long, cy As Long

Private Sub Form_DblClick()

    Unload Me

End Sub

Private Sub Form_Load()

    cx = 0

    cy = 0

    region(0) = CreateEllipticRgn(0, 0, 100, 100)

    bitmap = LoadImage(App.hInstance, App.Path & "\image.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

End Sub

Private Sub Form_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)

    Dim cx2 As Long, cy2 As Long

    cx2 = (X / 15) - 50

    cy2 = (Y / 15) - 50

    OffsetRgn region(0), cx2 - cx, cy2 - cy

    cx = cx2

    cy = cy2

    Me.Refresh

End Sub

Private Sub Form_Paint()

        SelectClipRgn hdc, region(0)

        Dim r As RECT

        GetClientRect hwnd, r

        dc = CreateCompatibleDC(hdc)

        SelectObject dc, bitmap

        StretchBlt hdc, 0, 0, r.Right, r.Bottom, dc, 0, 0, 800, 600, vbSrcCopy

        DeleteDC dc

End Sub

Private Sub Form_Resize()

    Me.Refresh

End Sub

Private Sub Form_Unload(Cancel As Integer)

    DeleteObject region(0)

    DeleteObject bitmap

End Sub

Después de haber arrancado el programa notaran que al mover el mouse se mueve un círculo que visualiza la imagen de fondo que hay en el formulario.

¿Cómo hacemos eso? En la sección de “caminos” vimos una función que permitía recortar una sección de una imagen que usábamos de fondo. Pues regiones también tiene una.

SelectClipRgn

Api: Declare Function SelectClipRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long) As Long

Esta función es la encargada de hacer ese recorte, se le pasa como primer parámetro el HDC que vamos a cortar y luego como segundo parámetro le pasamos la region. Hay algo curioso, veamos el evento Form_Paint:

        SelectClipRgn hdc, region(0)

        Dim r As RECT

        GetClientRect hwnd, r

        dc = CreateCompatibleDC(hdc)

        SelectObject dc, bitmap

        StretchBlt hdc, 0, 0, r.Right, r.Bottom, dc, 0, 0, 800, 600, vbSrcCopy

        DeleteDC dc

Aparte de las funciones GetClientRect y StretchBlt no hay nada nuevo, estas funciones serán explicadas en sus respectivas secciones, en general puede decirse que con GetClientRect obtenemos el tamaño del área cliente o dispositivo de contexto, y con StretchBlt ajustamos una imagen a un tamaño específico.

Lo que quería que vieran es que primero hay que hacer el SelectClipRgn antes de pegar la imagen, no se puede hacer al revés, ya que de hacerlo verán la imagen completa sin el recorte. Es esa la razón del por que no use la propiedad “Picture” del formulario, ya que al usarlo VB pinta primero la imagen y luego ejecuta Form_Paint por lo que no tenia los efectos esperados.

OffsetRgn

Api: Declare Function OffsetRgn Lib "gdi32" (ByVal hRgn As Long, ByVal X As Long, ByVal Y As Long) As Long

Una de los cosas curiosas es que cuando movemos el mouse se mueve la region, en realidad lo que estamos haciendo es “desplazando” la region, para ello usamos esta función, que tiene como primer parámetro la region a desplazar, y como segundo y tercer parámetro tenemos la cantidad de unidades lógicas a desplazar para el eje X e Y, respectivamente.

ExtSelectClipRgn

Api: Declare Function ExtSelectClipRgn Lib "gdi32" (ByVal hdc As Long, ByVal hRgn As Long, ByVal fnMode As Long) As Long

Esta función es casi igual a SelectClipRgn con la diferencia que permite aplicar las operaciones lógicas mencionadas en caminos, pero que vuelvo a repetir:

 Const RGN_AND = 1

 Const RGN_COPY = 5

 Const RGN_DIFF = 4

 Const RGN_OR = 2

 Const RGN_XOR = 3

Estas son las constantes para el parámetro “fnMode”, SelectClipRgn es lo equivalente a usar ExtSelectClipRgn con el parámetro “fnMode” = RGN_COPY.

Proyecto 3

Para este proyecto se necesita 2 botones, cuyo nombre sea “Command1” y “Command2” y una etiqueta o caption con nombre “label1” y la propiedad “BorderStyle” del formulario, ponerle el valor de “0- None”. Luego pegar este codigo:

Private Declare Function CreateEllipticRgn Lib "gdi32" (ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function CombineRgn Lib "gdi32" (ByVal hDestRgn As Long, ByVal hSrcRgn1 As Long, ByVal hSrcRgn2 As Long, ByVal nCombineMode As Long) As Long

Private Declare Function SetWindowRgn Lib "user32" (ByVal hwnd As Long, ByVal hRgn As Long, ByVal bRedraw As Boolean) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Dim region(0 To 4) As Long

Private Sub Command1_Click()

    Unload Me

End Sub

Private Sub Command2_Click()

    Unload Me

End Sub

Private Sub Form_Load()

    Command1.Left = 945

    Command1.Top = 765

    Command1.Width = 1050

    Command1.Height = 420

    Command2.Left = 1710

    Command2.Top = 1710

    Command2.Width = 1050

    Command2.Height = 330

    Label1.Left = 675

    Label1.Top = 315

    Label1.Width = 1635

    Label1.Height = 420

    Label1 = "Hola, al formulario eliptico"

    region(0) = CreateEllipticRgn(0, 0, 200, 100)

    region(1) = CreateEllipticRgn(100, 150, 200, 100)

    CombineRgn region(0), region(0), region(1), RGN_OR

    SetWindowRgn Me.hwnd, region(0), True

End Sub

Private Sub Form_Resize()

    Me.Refresh

End Sub

Private Sub Form_Unload(Cancel As Integer)

    DeleteObject region(0)

    DeleteObject region(1)

End Sub
Cuando arranquen el proyecto se sorprenderán que el formulario ya no tiene la forma rectangular tradicional sino que ha cambiado por  elipses, ¡sorprendente!, imaginen los formularios que a partir de ahora pueden crear, sin necesidad de continuar con el estilo rectangular tradicional, y si se ponen a ver los programas de hoy en día, tienen formas extrañas y originales. Aquí les digo como hacerlo. Como verán el código es simple, ya que hice algo sencillo, en realidad una aplicación completa requeriría de tiempo y de código.

La única función nueva es SetWindowRgn que explicamos a continuación.

SetWindowRgn

Api: Private Declare Function SetWindowRgn Lib "user32" (ByVal hwnd As Long, ByVal hRgn As Long, ByVal bRedraw As Boolean) As Long

Esta función es simple, como pueden ver esta función no toma el HDC del formulario sino directamente el HWND de la ventana. ¿Por qué se ven la elipses y no el rectángulo? Cuando ustedes crean un formulario, Windows le establece una region predeterminada en este caso la región rectangular, cuando el formulario se va a pintar pinta en base a la información de esta región, esta región abarca TODA la ventana, cuando me refiero a “TODO”, me refiero a la parte del CAPTION o barra de titulo, bordes, y el área cliente.

Es decir, esta función cambia el área de region que pinta la ventana, OJO ventana, no HDC, es decir, estamos hablando de todo el formulario. 

Como primer parámetro le pasamos el “hwnd” de la ventana, como segundo parámetro la región y como tercer parámetro le especificamos si repinta la ventana o no. Si se coloca TRUE inmediatamente se envía un mensaje para pintar el formulario, en caso de pasarle FALSE el formulario no se pinta inmediatamente.

Noten, que un efecto que pudimos haber considerado complicado, vean lo sencillo que es. A ustedes les dejo, realizar el código de MOVER el formulario, como pueden ver el formulario se queda estático y no es posible moverlo, pero echando unas líneas de código, de seguro no tendrán problema para hacerlo. 

Proyecto 4

En el proyecto 3 quedamos seguramente sorprendidos al ver que podíamos crear un formulario con forma irregular, pues en este proyecto tal vez queden aun mas sorprendido al ver la versatilidad de la creación de este tipo de formularios. Puede que para algunas aplicaciones sea complicado y engorroso la creación de regiones para crear un formulario de manera irregular.

Imaginen que tiene una imagen, y tiene como fondo un color que denominamos “color de transparencia”. Y también imaginen que pudiéramos quitar el color de transparencia dejando nada más la imagen que nos interesa.

Pues en este proyecto hacemos eso con la utilización de regiones. Para este proyecto es necesario un formulario con la propiedad BorderStyle = “0-None”, un botón cuyo nombre sea “Command1” y que la imagen “imagen3.bmp” que acompaña a este manual, sea colocada en el lugar donde esta el proyecto.

Luego ponen este código en el formulario:

Const LR_LOADFROMFILE = &H10

Const IMAGE_BITMAP = 0

Private Type POINTAPI

        X As Long

        Y As Long

End Type

Private Type RECT

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

Const RGN_OR = 2

Private Declare Function CombineRgn Lib "gdi32" (ByVal hDestRgn As Long, ByVal hSrcRgn1 As Long, ByVal hSrcRgn2 As Long, ByVal nCombineMode As Long) As Long

Private Declare Function SetWindowRgn Lib "user32" (ByVal hwnd As Long, ByVal hRgn As Long, ByVal bRedraw As Boolean) As Long

Private Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal dwImageType As Long, ByVal dwDesiredWidth As Long, ByVal dwDesiredHeight As Long, ByVal dwFlags As Long) As Long

Private Declare Function CreateRectRgn Lib "gdi32" (ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function BitBlt Lib "gdi32" (ByVal hDestDC As Long, ByVal X As Long, ByVal Y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal dwRop As Long) As Long

Private Declare Function DeleteDC Lib "gdi32" (ByVal hdc As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Declare Function CreateCompatibleDC Lib "gdi32" (ByVal hdc As Long) As Long

Private Declare Function GetPixel Lib "gdi32" (ByVal hdc As Long, ByVal X As Long, ByVal Y As Long) As Long

Dim region(0 To 4) As Long

Dim bitmap As Long

Dim dc As Long

Dim primeravez As Boolean

Private Sub Command1_Click()

    Unload Me

End Sub

Private Sub Form_Load()

    Command1.Left = 1295

    Command1.Top = 1440

    Command1.Height = 645

    Command1.Width = 1275

    Command1.Caption = "Salir"

    bitmap = LoadImage(App.hInstance, App.Path & "\imagen3.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

    dc = CreateCompatibleDC(Me.hdc)

    SelectObject dc, bitmap

    primeravez = True

    ChequeoPixeles

    Me.Width = 260 * 15

    Me.Height = 250 * 15

    SetWindowRgn Me.hwnd, region(0), True

End Sub

Sub ChequeoPixeles()

    Dim i As Long, j As Long

    Dim pixel As Long

    For i = 1 To 259

        For j = 1 To 219

            pixel = GetPixel(dc, i, j)

            If pixel <> RGB(255, 255, 0) Then

                If primeravez Then

                    region(0) = CreateRectRgn(i, j, i + 1, j + 1)

                    primeravez = False

                Else

                    region(1) = CreateRectRgn(i, j, i + 1, j + 1)

                    CombineRgn region(0), region(0), region(1), RGN_OR

                    DeleteObject region(1)

                End If

            End If

        Next

    Next

End Sub

Private Sub Form_Paint()

        BitBlt hdc, 0, 0, 260, 220, dc, 0, 0, vbSrcCopy

End Sub

Private Sub Form_Unload(Cancel As Integer)

    DeleteObject region(0)

    DeleteObject region(1)

    DeleteDC dc

    DeleteObject bitmap

End Sub

Private Sub Form_Resize()

    Me.Refresh

End Sub

Al correr el proyecto, se sorprenderán al ver un formulario con una forma no tradicional, hecha a partir de una imagen, si ven la imagen por un programa grafico notaran que tiene fondo amarillo, el amarillo es mi “color de transparencia” le digo así ya que en realidad lo que vamos hacer es crear regiones a partir de los píxeles que no son amarillos.

¿Cómo lo hacemos? Veamos la función encargada de crear la región, cabe mencionar que la imagen tiene una medida de 260 X 220:

Sub ChequeoPixeles()

    Dim i As Long, j As Long

    Dim pixel As Long

    For i = 1 To 259 ‘ Verificamos las 259 filas que tiene la imagen
        For j = 1 To 219 ‘ Verificamos las 219 columnas de cada fila

            pixel = GetPixel(dc, i, j) ‘ Obtenemos el color de 1 pixel en especifico

            If pixel <> RGB(255, 255, 0) Then ‘ Si el píxel es distinto de amarillo

                If primeravez Then ‘ ¿Es primera vez que entra?
                    region(0) = CreateRectRgn(i, j, i + 1, j + 1) ‘ Si es primera vez entonces almacenamos la primera region en region(0)

                    primeravez = False ‘ Lo ponemos false para que no vuelva a entrar
                Else ‘ Si no es primera vez
                    region(1) = CreateRectRgn(i, j, i + 1, j + 1) ‘ La siguiente region la almacenamos en region(1)

                    CombineRgn region(0), region(0), region(1), RGN_OR ‘ Combinamos las dos regiones

                    DeleteObject region(1) ‘ Eliminamos region(1) ya que no es útil en este momento

                End If

            End If

        Next ‘ Continuamos hasta chequear todas las columnas de una fila
    Next ‘ Continuamos hasta chequear todas las filas
End Sub

Creo que los comentarios en azul explican mas o menos con detalle lo que hace la función, básicamente lo que estoy haciendo es crear regiones de 1 X 1 píxel, por cada píxel que no es de color amarillo, acumulo todas esas regiones cuando uso CombineRgn, y luego al final de todo, region(0) tiene todas las regiones o véanlo así, todos los píxeles que no son de color amarillo. Luego con la función SetWindowRgn, establecemos nuestra nueva region.

Dejo en claro, que el código es un poco rudimentario, en el sentido que si la imagen es muy grande se puede hacer lento el arranque del programa, en Internet y en libros pueden encontrar formas de optimizar este código, aunque dudo mucho que consigan el código en VB. A ustedes les dejo la optimización de todo esto. Recuerden que yo les doy la idea  y las herramientas. Ustedes, se sientan a mejorar y optimizar todo esto,  personalmente yo no he trabajado mucho con esto, ya que no se me ha dado el momento. Pero de seguro que muchas personas habrán visto interesante todo esto. Y estoy seguro que por sus cabezas pasan millones de ideas de cómo crear una aplicación, que antes no podían hacer.

Funciones Extras

EqualRgn

Api: Declare Function EqualRgn Lib "gdi32" (ByVal hSrcRgn1 As Long, ByVal hSrcRgn2 As Long) As Long

Función utilizada para ver si dos regiones son iguales. Esta función retorna TRUE si son iguales y FALSE en caso de no serlo. Estas dos regiones vienen dadas por los dos únicos parámetro de la función.

RectInRegion

Api: Declare Function RectInRegion Lib "gdi32" (ByVal hRgn As Long, lpRect As RECT) As Long

Función que sirve para ver si un rectángulo, cuyas coordenadas se pasan en el segundo parámetro, que es de tipo RECT, se encuentra dentro de la región especificada en el primer parámetro. Al igual que la anterior esta función devuelve TRUE o FALSE

GetRgnBox

Api: Declare Function GetRgnBox Lib "gdi32" (ByVal hRgn As Long, lpRect As RECT) As Long

Algunas veces tenemos regiones irregulares, pero queremos saber el rectángulo mínimo que enmarca la región, pues esta función es la encargada de ello, como primer parámetro si le pasa la region a enmarcar y como segundo parámetro se le pasa una estructura de tipo RECT la cual al finalizar la función, tendrá las coordenadas del rectángulo mínimo que enmarca la región.

Esto es todo acerca de regiones, si quieren ver todas las funciones API’S referente a este tema vean la documentación oficial de Microsoft en Internet, yo nada mas introduje las funciones que considere como las mas importantes. A partir de ahora estoy seguro que tienen otra visión para crear sus proyectos, con efectos especiales muy buenos y fáciles de hacer.

FUENTES

Vamos a introducir un tema que es relativamente complejo, ya que las funciones que manejan o manipulan las fuentes están llenas de muchos detalles y muchas variables, inclusive en esta parte no soy muy experto así que me voy a limitar a explicar lo que conozco. Ya que hay algunas constantes que en realidad no entiendo muy bien.

¿Por qué estudiar las fuentes? Windows proporciona unas funciones muy poderosas para la manipulación de fuentes, particularmente yo nunca las he usado, pero algunas veces si necesitamos escribir un texto directamente en el HDC, pues será necesario conocer como funcionan estas funciones, ya que tal vez los controles en este caso no nos sirvan.

Antes de explicar la parte teórica de todo esto, vamos a decir que es un “serif”, para empezar no es lo que yo creía, en particular ya pensaba que era un tipo de letra, pero no es exactamente eso, para explicar lo que es, veamos una letra Times New Roman y otra Arial:

T T
Al lado izquierdo tengo la T de Times New Roman y del lado derecho tengo la T de arial. Vean las puntas de la T de Times New Roman, hay como una pequeña silueta o triangulito al final de cada extremo, esa silueta particular es lo que se llama Serif, es decir, Times New Roman es una fuente Serif, se dice así a las fuentes que tienen cierta extensión en los extremos. Arial por ejemplo, no es serif ya que carece de esa extensión, como pueden ver sus extremos son completamente rectangulares. Otro ejemplo de una fuente no serif es la Courier New, cuya T:

T
Noten que aquí el extremo tiene como una extensión, pero esa extensión sigue siendo recto!, es decir, como rectangular. Esto lo explico ya que ahora me toca explicar los diferentes tipos de familias que conforman a Windows.

Familias

DECORATIVE

Valor: FF_DECORATIVE = 80 

Son esos estilos de fuentes sofisticados, empleados en esos documentos de estilo medieval.

DONTCARE

Valor: FF_DONTCARE = 0    

Especifica una familia genérica. Este se utiliza cuando la información acerca de la fuente no importa o no existe. Y se emplea la fuente por defecto.

MODERN

Valor:  FF_MODERN = 48

Son fuentes cuyo ancho en todos los caracteres son iguales, y no son serif. Ejemplo Courire New:

ABCDEFGHI, pueden ver a simple vista que en la cadena mostrada, cada letra tiene la misma anchura.

ROMAN

Valor: FF_ROMAN = 16 

Son fuentes cuyo ancho de los caracteres son variables, y son Serif. Ejemplo: Times New Roman.

SCRIPT

Valor: FF_SCRIPT = 64 

Son fuentes que se asemeja mucho a la escritura humana.

SWISS

Valor: FF_SWISS = 32

En esta familia están contenidas las fuentes cuyo ancho de caracteres son variables pero no son Serif, por ejemplo: Arial.

Tipos de Fuente

Raster:

En este tipo de fuente los caracteres vienen representados por BITMAP, que el sistema usa para pintar cada carácter.

Vector:

Aquí los caracteres son una sería de líneas que definen segmentos, que el sistema usa para dibujar el carácter.

TrueType y OpenType:

Es una colección de instrucciones que dibujan líneas y curvas, dando forma a cada letra.

Grupo de caracteres

La mayoría debe de conocer que todos los caracteres no son más que números, que son identificados por Windows y se le asigna un carácter dependiendo de su código. Por ejemplo es clásico saber que para la letra “ñ” si presionamos ALT-164, se muestra en pantalla la “ñ” pequeña, el código es 164, eso significa que existe un grupo de caracteres con cierto código y dependiendo del código se muestra el carácter. Existen varios grupos de caracteres que voy a explicarlos brevemente:

Grupo de caracteres “Windows”.

Este el mas usado consta de 8 bit, es decir, cada carácter es un byte, este va desde ​0x20 (decimal 32) hasta 0xFF (255), es decir, tenemos 132 caracteres disponibles, este grupo empieza con el carácter en blanco (“ “). Si lo quieren ver aprieten ALT-32.

Grupo de caracteres “UNICODE”.

El grupo de caracteres “Windows”, funciona perfecto para algunos lenguajes, pero existía un problema, cuando esto de la computación se globaliza a nivel mundial, nació la necesidad de hacer compatibles los software con cada lenguaje, el problema es que existen lenguajes con mas de 255 caracteres, esto obligo a los programadores, a crear un nuevo grupo de caracteres que se llama Unicode. Este grupo esta formado por 16 bit (2 byte), lo que permite 65,536 caracteres, en los cuales se incluye símbolos matemáticos, signos de puntuación ,etc.

Grupo de caracteres “OEM”.

Este grupo es el usado en el sistema MS-DOS®. En este grupo desde los caracteres 32 hasta el 127 son idénticos a los del grupo ASCII, y Windows. 

Grupo de caracteres “símbolos”.

Grupo que contiene símbolos de uso matemático y científico.

Dimensiones

Cuando vimos las familias mencione que algunas consistían de grupo de caracteres de tamaño variable. Eso significa que de alguna manera, para hacer una aplicación que manipula letras, deberíamos de conocer como son esas dimensiones. Esto en realidad es todo un problema, ya que hay muchas medidas, en realidad como vamos a ver mas adelante existe una estructura TEXTMETRIC el cual contiene esas medidas, evidentemente contendrá las medidas de la fuente que este configurada en nuestro dispositivo de contexto.

La estructura es así:

Private Type TEXTMETRIC

    tmHeight As Long  / Altura de la letra o fuente

    tmAscent As Long 

    tmDescent As Long

    tmInternalLeading As Long

    tmExternalLeading As Long

    tmAveCharWidth As Long / Anchura media

    tmMaxCharWidth As Long / Anchura Maxima
    tmWeight As Long / Grosor o Peso de la letra o fuente

    tmOverhang As Long

    tmDigitizedAspectX As Long

    tmDigitizedAspectY As Long

    tmFirstChar As Byte / Primer carácter de la fuente
    tmLastChar As Byte / Último carácter de la fuente

    tmDefaultChar As Byte / carácter por defecto

    tmBreakChar As Byte

    tmItalic As Byte / Indica si la letra esta cursiva

    tmUnderlined As Byte / Indica si la letra esta subrayada

    tmStruckOut As Byte / Indica si la letra esta tachada
    tmPitchAndFamily As Byte / Indica a la familia a la cual pertenece
    tmCharSet As Byte /  Indica el grupo de caracteres al cual pertenece
End Type

Algunas de los parámetros los explico de manera grafica y para ello voy a tomar la imagen que se encuentra en la documentación oficial de MSDN:

Imagen tomada de la página de MSDN en Internet.

En este caso, otmMacAscent es igual a tmAscent, y otmMacDescent es igual a tmDescent. Lo que sucede es que las variables que empiezan con “otm” pertenecen a otra estructura mas completa que es NEWTEXTMETRIC, esta estructura tiene todos los parámetros de TEXTMETRIC y unos adicionales, los parámetros que no explico en realidad no los entiendo, tal vez por que todavía no le he visto utilidad a esos parámetros.

¿Para que todo esto? Si ustedes quisieran diseñar un programa parecido a WORD tendrían que conocer muy bien esto, por que con esto controlan la posición de cada carácter en el momento de dibujar. En el ejemplo que mostrare a continuación yo no trabaje esta parte, ya que esto requiere de cierto trabajo, que en realidad no lo veo mucha utilidad ya que existen controles que hacen ese trabajo. Pero uno nunca sabe! Por ejemplo, si por alguna circunstancia necesitan hacer su propio TEXTBOX le aseguro que tendría que usar todo lo hablado con anterioridad.

Vayamos al proyecto, para este programa es necesario los siguientes controles: 3 ListBox y 3 etiquetas o Label y un modulo. Luego pongan este código:

En el formulario pongan este código:

Const LOGPIXELSY = 90

Const DEFAULT_CHARSET = 1

Const OUT_DEFAULT_PRECIS = 0

Const CLIP_DEFAULT_PRECIS = 0

Const DEFAULT_QUALITY = 0

Const DEFAULT_PITCH = 0

Const DT_BOTTOM = &H8

Const OPAQUE = 2

Const TRANSPARENT = 1

Const DT_CENTER = &H1

Const DT_WORDBREAK = &H10

Const DT_END_ELLIPSIS = &H8000

Const DT_MODIFYSTRING = &H10000

Const DT_PATH_ELLIPSIS = &H4000

Const DT_TOP = &H0

Const ANTIALIASED_QUALITY = 4

Private Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Sub Form_Load()

    List1.Top = 90

    List1.Left = 135

    List1.Width = 3165

    List1.Height = 2400

    List2.Top = 90

    List2.Left = 3375

    List2.Width = 3210

    List2.Height = 2400

    List3.Top = 90

    List3.Left = 6660

    List3.Width = 3210

    List3.Height = 2400

    Label1.Top = 2520

    Label1.Left = 135

    Label1.Width = 3165

    Label1.Height = 420

    Label2.Top = 2520

    Label2.Left = 3375

    Label2.Width = 3210

    Label2.Height = 420

    Label3.Top = 2520

    Label3.Left = 6660

    Label3.Width = 3210

    Label3.Height = 420

    Me.Width = 10080

    Me.Height = 6780

    EnumFontFamilies Me.hdc, vbNullString, AddressOf FamiliaLetras, ByVal 0&

    Label1 = "Cantidad de letras RASTER = " & cantidad(1)

    Label2 = "Cantidad de letras TRUETYPE = " & cantidad(2)

    Label3 = "Cantidad de letras DEVICE = " & cantidad(3)

End Sub

Private Sub List1_Click()

    CrearLetra List1.Text

End Sub

Private Sub List2_Click()

    CrearLetra List2.Text

End Sub

Private Sub List3_Click()

    CrearLetra List3.Text

End Sub

Sub CrearLetra(letra As String)

    Me.Cls

    Dim fuente As Long

    fuente = CreateFont(-MulDiv(30, GetDeviceCaps(Me.hdc, LOGPIXELSY), 72), 0, 0, 0, 0, 0, 0, 0, DEFAULT_CHARSET, OUT_DEFAULT_PRECIS, CLIP_DEFUALT_PRECIS, ANTIALIASED_QUALITY, DEFAULT_PITCH, letra)

    Dim cadena As String

    Dim r As RECT

    Dim s As Size

    r.Left = 10

    r.Top = 225

    r.Bottom = 400

    r.Right = 600

    SelectObject Me.hdc, fuente

    Rectangle Me.hdc, 10, 225, 600, 400

    'Comentario 1

    'SetBkMode Me.hdc, OPAQUE

    'SetBkColor Me.hdc, RGB(0, 0, 255)

    'SetTextColor Me.hdc, RGB(255, 255, 255)    

    'Comentario 2

    'cadena = "Hola Mundo!"

    'GetTextExtentPoint32 Me.hdc, cadena, Len(cadena), s

    'Rectangle Me.hdc, 10, 225, 10 + s.cx, 225 + s.cy

    'DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP

    'Comentario 3

    'cadena = "Hola! esta es una cadena muyyyy larga"

    'DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP Or DT_CENTER Or DT_WORDBREAK

    'Comentario 4

    'cadena = "Hola! esta es una cadena muyyyy larga"

    'DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP Or DT_CENTER Or DT_END_ELLIPSIS Or DT_MODIFYSTRING

    'Comentario 5

    'cadena = "Hola otra vez Mundo!"

    'GetTextExtentPoint32 Me.hdc, cadena, Len(cadena), s

    'Rectangle Me.hdc, 10, 225, 10 + s.cx, 225 + s.cy

    'SetTextJustification Me.hdc, 600 - s.cx, 4

    'DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP

    'Comentario 6

    'cadena = "Hola Mundo!"

    'SetTextCharacterExtra Me.hdc, 30

    'DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP

    'Comentario 7

    cadena = "Hola Mundo!"

    DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP        

    DeleteObject fuente

    Me.Refresh

End Sub
En el modulo:

Public Const LF_FULLFACESIZE = 64

Public Const LF_FACESIZE = 32

Public Const DEVICE_FONTTYPE = &H2

Public Const RASTER_FONTTYPE = &H1

Public Const TRUETYPE_FONTTYPE = &H4

Public Type Size

        cx As Long

        cy As Long

End Type

Public Type RECT

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

Public Type TEXTMETRIC

        tmHeight As Long

        tmAscent As Long

        tmDescent As Long

        tmInternalLeading As Long

        tmExternalLeading As Long

        tmAveCharWidth As Long

        tmMaxCharWidth As Long

        tmWeight As Long

        tmOverhang As Long

        tmDigitizedAspectX As Long

        tmDigitizedAspectY As Long

        tmFirstChar As Byte

        tmLastChar As Byte

        tmDefaultChar As Byte

        tmBreakChar As Byte

        tmItalic As Byte

        tmUnderlined As Byte

        tmStruckOut As Byte

        tmPitchAndFamily As Byte

        tmCharSet As Byte

End Type

Public Type LOGFONT

        lfHeight As Long

        lfWidth As Long

        lfEscapement As Long

        lfOrientation As Long

        lfWeight As Long

        lfItalic As Byte

        lfUnderline As Byte

        lfStrikeOut As Byte

        lfCharSet As Byte

        lfOutPrecision As Byte

        lfClipPrecision As Byte

        lfQuality As Byte

        lfPitchAndFamily As Byte

        lfFaceName(1 To LF_FACESIZE) As Byte

End Type

Public Type ENUMLOGFONT

        elfLogFont As LOGFONT

        elfFullName(LF_FULLFACESIZE) As Byte

        elfStyle(LF_FACESIZE) As Byte

End Type

Public Type NEWTEXTMETRIC

        tmHeight As Long

        tmAscent As Long

        tmDescent As Long

        tmInternalLeading As Long

        tmExternalLeading As Long

        tmAveCharWidth As Long

        tmMaxCharWidth As Long

        tmWeight As Long

        tmOverhang As Long

        tmDigitizedAspectX As Long

        tmDigitizedAspectY As Long

        tmFirstChar As Byte

        tmLastChar As Byte

        tmDefaultChar As Byte

        tmBreakChar As Byte

        tmItalic As Byte

        tmUnderlined As Byte

        tmStruckOut As Byte

        tmPitchAndFamily As Byte

        tmCharSet As Byte

        ntmFlags As Long

        ntmSizeEM As Long

        ntmCellHeight As Long

        ntmAveWidth As Long

End Type

Declare Function EnumFontFamilies Lib "gdi32" Alias "EnumFontFamiliesA" (ByVal hdc As Long, ByVal lpszFamily As String, ByVal lpEnumFontFamProc As Long, ByVal lParam As Long) As Long

Declare Function SetBkColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long

Declare Function SetBkMode Lib "gdi32" (ByVal hdc As Long, ByVal nBkMode As Long) As Long

Declare Function SetTextAlign Lib "gdi32" (ByVal hdc As Long, ByVal wFlags As Long) As Long

Declare Function SetTextCharacterExtra Lib "gdi32" (ByVal hdc As Long, ByVal nCharExtra As Long) As Long

Declare Function SetTextColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long

Declare Function SetTextJustification Lib "gdi32" (ByVal hdc As Long, ByVal nBreakExtra As Long, ByVal nBreakCount As Long) As Long

Declare Function GetTextAlign Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function GetTextCharacterExtra Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function GetTextExtentPoint32 Lib "gdi32" Alias "GetTextExtentPoint32A" (ByVal hdc As Long, ByVal lpsz As String, ByVal cbString As Long, lpSize As Size) As Long

Declare Function GetTextColor Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function GetBkColor Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function GetBkMode Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function CreateFont Lib "gdi32" Alias "CreateFontA" (ByVal H As Long, ByVal W As Long, ByVal E As Long, ByVal O As Long, ByVal W As Long, ByVal I As Long, ByVal u As Long, ByVal s As Long, ByVal C As Long, ByVal OP As Long, ByVal CP As Long, ByVal Q As Long, ByVal PAF As Long, ByVal F As String) As Long

Declare Function GetTextMetrics Lib "gdi32" Alias "GetTextMetricsA" (ByVal hdc As Long, lpMetrics As TEXTMETRIC) As Long

Declare Function CreateFontIndirect Lib "gdi32" Alias "CreateFontIndirectA" (lpLogFont As LOGFONT) As Long

Declare Function MulDiv Lib "kernel32" (ByVal nNumber As Long, ByVal nNumerator As Long, ByVal nDenominator As Long) As Long

Declare Function DrawText Lib "user32" Alias "DrawTextA" (ByVal hdc As Long, ByVal lpStr As String, ByVal nCount As Long, lpRect As RECT, ByVal wFormat As Long) As Long

Declare Function GetDeviceCaps Lib "gdi32" (ByVal hdc As Long, ByVal nIndex As Long) As Long

Public cantidad(1 To 3) As Long

Public infLetra As TEXTMETRIC

Public cnfLetra As LOGFONT

Function FamiliaLetras(lpNLF As ENUMLOGFONT, lpNTM As NEWTEXTMETRIC, ByVal FontType As Long, lParam As Long) As Long

    Dim FaceName As String, Familia As String

    FaceName = StrConv(lpNLF.elfLogFont.lfFaceName, vbUnicode)

    If FontType And RASTER_FONTTYPE Then

        frmfamilias.List1.AddItem Left$(FaceName, InStr(FaceName, vbNullChar) - 1)

        cantidad(1) = cantidad(1) + 1

    End If

    If FontType And TRUETYPE_FONTTYPE Then

        frmfamilias.List2.AddItem Left$(FaceName, InStr(FaceName, vbNullChar) - 1)

        cantidad(2) = cantidad(2) + 1

    End If

    If FontType And DEVICE_FONTTYPE Then

        frmfamilias.List3.AddItem Left$(FaceName, InStr(FaceName, vbNullChar) - 1)

        cantidad(3) = cantidad(3) + 1

    End If

    FamiliaLetras = 1

End Function

Debo mencionar que hay varias funciones encargadas de presentar texto en un HDC, particularmente voy a explicar la que yo considero mas completa y que cubre todo lo que hacen las demás funciones, hablo de la función DrawText.

Al arrancar el proyecto lo primero que vemos son 3 listbox, uno de ellos contiene las fuentes de tipo Raster, el otro las fuentes del tipo TrueType, y el último las fuentes de dispositivo, particularmente yo, no tenia ninguna fuente en el tercer listbox, es decir, las fuentes de dispositivo. También notaran que a medida que pasean por las fuentes es mostrado en la parte inferior un texto “Hola Mundo!”, con la fuente seleccionada. Empecemos explicando la función encargada de retornar la familia de fuentes instaladas en su computadora.

EnumFontFamilies

Api: Declare Function EnumFontFamilies Lib "gdi32" Alias "EnumFontFamiliesA" (ByVal hdc As Long, ByVal lpszFamily As String, ByVal lpEnumFontFamProc As Long, ByVal lParam As Long) As Long

Esta función se encarga de retornarnos las fuentes instaladas en nuestro sistema, existe otra función que es EnumFonts, pero esta que explicamos hace lo mismo y es un poco mas completa. 

Como primer parámetro le pasamos el HDC, como segundo parámetro podemos filtrar las fuentes en base a la familia a la cual pertenecen, como tercer parámetro le pasamos la dirección de una función y por último podemos pasarle alguna información que nosotros creamos conveniente.

Este tipo de funciones es raro para ustedes, ya que es la primera vez que usamos una de este estilo. ¿Qué quiero decir?. Vean que como parámetro le pasamos la dirección de una función, eso nada mas lo habíamos hecho cuando cambiamos nuestro gestor de mensaje, en la sección de Sistema de coordenadas. Ustedes se preguntaran ¿Para que una función?. Vayamos al ejemplo:

EnumFontFamilies Me.hdc, vbNullString, AddressOf FamiliaLetras, ByVal 0&

Le pasamos nuestro HDC, indicamos “vbnullstring” en el segundo parámetro y con esto le indicamos que queremos a todas las familias de fuentes, como tercer parámetro le pasamos la dirección de la función que se encuentra en el modulo, y como no quiero pasar ninguna información adicional, le paso 0 al último parámetro.

Veamos la función en el modulo:

Function FamiliaLetras(lpNLF As ENUMLOGFONT, lpNTM As NEWTEXTMETRIC, ByVal FontType As Long, lParam As Long) As Long

    Dim FaceName As String, Familia As String

    FaceName = StrConv(lpNLF.elfLogFont.lfFaceName, vbUnicode)

    If FontType And RASTER_FONTTYPE Then

        frmfamilias.List1.AddItem Left$(FaceName, InStr(FaceName, vbNullChar) - 1)

        cantidad(1) = cantidad(1) + 1

    End If

    .

    .

    .

    .

    FamiliaLetras = 1

End Function

Esta función tiene que cumplir cierta estructura, es decir, debe de recibir los parámetros especificados en el enunciado de dicha función. Como primer parámetro tenemos una variable de tipo ENUMLOGFONT, como segundo parámetro tenemos una variables de tipo NEWTEXTMETRIC, que es casi igual a TEXTMETRIC, tenemos como tercer parámetro una variable que nos informa del tipo de fuente que esta siendo pasada, y por último tenemos la información adicional que pudimos haber pasado cuando invocamos a la función EnumFontFamilies.

Es importante ver que hacemos FamiliaLetras = 1, con esto le decimos a Windows que todavía hay fuentes por verificar, es decir, lo que hace el programa es entrar a esta función tantas veces como fuentes existan. Por cada fuente que exista, Windows invocara a dicha función.

¿Qué hacemos adentro de la función?. En primer lugar tomamos el nombre de la fuente, eso lo hacemos con:

    FaceName = StrConv(lpNLF.elfLogFont.lfFaceName, vbUnicode)
De la estructura ENUMLOGFONT, el parámetro que yo considero mas importante es la variable que es de tipo LOGFONT, los variables contenidas en esta estructura las explicare cuando explique la función CreateFont. Pero la única variable que nos interesa ahora es lfFaceName, la cual contiene el nombre de la fuente. ¿Por qué usar StrConv?. StrConv es una función de VB, si ven de que tipo es lfFaceName, se darán cuenta que un arreglo de Byte. Lo que hace StrConv, es convertir ese arreglo de Byte, en una cadena de texto completamente valida que se pueda almacenar en FaceName que es tipo String.

Después vemos un if:

    If FontType And RASTER_FONTTYPE Then

Y con esto lo que hago es separar las fuentes por su tipo. Llenando como pueden ver los respectivos ListBox en los 3 IF que tiene la función.

Teniendo el nombre de las fuentes, ahora vamos ver las funciones necesarias para manipularlas.

CreateFont

Api: Declare Function CreateFont Lib "gdi32" Alias "CreateFontA" (ByVal nHeight As Long, ByVal nWidth As Long, ByVal nEscapement As Long, ByVal nOrientation As Long, ByVal nWeight As Long, ByVal Italic As Long, ByVal underline As Long, ByVal strikeout As Long, ByVal Charset As Long, ByVal OutPutPrecision As Long, ByVal ClipPrecicion As Long, ByVal Quality As Long, ByVal PitchAndFamily As Long, ByVal Face As String) As Long

Antes que nada esta función no es para crear una fuente nueva!. Esta función lo que hace es crear un objeto tipo fuente, cosa muy distinta. Es decir, se manipula como un objeto, igual como hacíamos con la brocha y pincel. Como pueden ver los parámetros son los mismos que las variables que se encuentran en la estructura LOGFONT. Por consiguiente la explicación de estos parámetros se aplica perfectamente a la estructura LOGFONT.

nHeight = Este parámetro establece la altura de la fuente. Pero dependiendo del valor que tenga se puede comportar de manera distinta.

Si es igual 0, el sistema toma la altura de la fuente por defecto.

Si es mayor que 0,  la altura será la medida tmHeight, mostrada en la figura de arriba.

Si es menor que 0, la altura será la suma de tmHeight + tmInternalLeading.

Esta altura se establece en unidades lógica. Si trabajamos con MM_TEXT que es lo equivalente a trabajar con píxeles, la altura viene dada por la siguiente formula:

MulDiv(30, GetDeviceCaps(Me.hdc, LOGPIXELSY), 72)

Si quieren alterar el tamaño, modifiquen el primer parámetro de la función MulDiv, es decir, donde esta el 30. Al final explico estas dos funciones.

nWidth = determina el ancho de la fuente.

nEscapement = Aquí especificamos un ángulo de rotación, es decir, no  necesariamente el texto se tiene que presentar de forma horizontal, el texto o carácter puede presentarse con cierto grado de inclinación. Ejm



Este parámetro toma el ángulo de rotación, si el valor es positivo se toma la rotación en sentido contrario a las agujas del reloj, en caso de ser negativo la rotación se realiza en sentido de las agujas del reloj. Los grados que vayan a pasar deben de ser multiplicado por 10, es decir, si queremos rotar 45 grados, tienen que colocar 450.

nOrientation = este parámetro solo funciona en Windows NT/2000, es un ángulo de giro, pero por carácter. Y antes de crear la fuente hay que llamar la función SetGraphicsMode, pasándole el parámetro GM_ADVANCED. Ejm:


Pueden ver como los caracteres están rotados o girados en cierto grado.

nWeight = Determina el grosor o peso de la fuente, es decir, si quieren poner en negritas un texto esta es la variables donde especifican la intensidad. Los valores a tomar pueden ser:

Private Const FW_BLACK = FW_HEAVY

Private Const FW_BOLD = 700

Private Const FW_DEMIBOLD = FW_SEMIBOLD

Private Const FW_DONTCARE = 0

Private Const FW_EXTRABOLD = 800

Private Const FW_EXTRALIGHT = 200

Private Const FW_HEAVY = 900

Private Const FW_LIGHT = 300

Private Const FW_MEDIUM = 500

Private Const FW_NORMAL = 400

Private Const FW_REGULAR = FW_NORMAL

Private Const FW_SEMIBOLD = 600

Private Const FW_THIN = 100

Private Const FW_ULTRABOLD = FW_EXTRABOLD

Private Const FW_ULTRALIGHT = FW_EXTRALIGHT

Italic = determina si la fuente se crea “cursiva”, si se coloca 1 la fuente será cursiva, en caso contrario deberán poner 0.

Underline = determina si la fuente estará “subrayada”, al igual que Italic el colocar 1 implica activar la propiedad y 0 no tomarla en cuenta.

StrikeOut = determina si la fuente estará “tachada”, al igual que Italic el colocar 1 implica activar la propiedad y 0 no tomarla en cuenta.

Charset = en este parámetro hablamos de los “grupos de caracteres” que explicamos arriba, es decir, algunos caracteres pueden tener diferentes representación dependiendo de la familia a la cual pertenece, por ejemplo, puede existir una Arial (Alemán) y una Arial (China), la letra “A” tiene distintas formas de representarse, pero ambas son Ariales, pero se representan de manera distinta. Los valores que puede tomar este parámetro son:

Private Const ANSI_CHARSET = 0

Private Const BALTIC_CHARSET = 186

Private Const CHINESEBIG5_CHARSET = 136

Private Const DEFAULT_CHARSET = 1

Private Const EASTEUROPE_CHARSET = 238

Private Const GB2312_CHARSET = 134

Private Const GREEK_CHARSET = 161

Private Const HANGUL_CHARSET = 129

Private Const HANGEUL_CHARSET = 129

Private Const MAC_CHARSET = 77

Private Const OEM_CHARSET = 255

Private Const RUSSIAN_CHARSET = 204

Private Const SHIFTJIS_CHARSET = 128

Private Const SYMBOL_CHARSET = 2

Private Const TURKISH_CHARSET = 162

OutPutPrecision =  Aquí especificamos que tipo de fuente se quiere tomar, es decir, podemos tener una fuente Arial de tipo TrueType y otra Arial del tipo Raster, con este parámetro especificamos cual de las 2 queremos tomar, ya que ambas serían Arial, es decir, tienen el mismo nombre. Los valores que puede tomar son:

Private Const OUT_DEVICE_PRECIS = 5

Private Const OUT_DEFAULT_PRECIS = 0

Private Const OUT_OUTLINE_PRECIS = 8

Private Const OUT_RASTER_PRECIS = 6

Private Const OUT_TT_ONLY_PRECIS = 7

Private Const OUT_TT_PRECIS = 4

ClipPresicion = Aquí se especifica el como se recortaran los caracteres cuando la cadena sea mas grande que el área de recorte. Este es uno de los parámetros que todavía no entiendo bien lo que hace, pero como valores puede tomar:

Private Const CLIP_DEFAULT_PRECIS = 0

Private Const CLIP_CHARACTER_PRECIS = 1

Private Const CLIP_EMBEDDED = 128

Private Const CLIP_MASK = &HF

Private Const CLIP_STROKE_PRECIS = 2

Private Const CLIP_LH_ANGLES = 16

Private Const CLIP_TT_ALWAYS = 32

Quality = Con este parámetro establecemos la calidad de la letra, es decir, si ven una letra con un tamaño grande, verán que las líneas son como rectángulos uno encima de otro, con este parámetro podemos hacer lo que se llama ANTIALIASED, en donde esos rectángulos son manipulados para parecerse mas a una línea completamente recta, logrando así una calidad mucho mejor en el momento de mostrar los caracteres. Los valores que puede tomar:

Private Const ANTIALIASED_QUALITY = 4 ‘ Mejora la calidad de presentación

Private Const DEFAULT_QUALITY = 0 ‘ Dejamos a Windows que tome valores por defecto

Private Const DRAFT_QUALITY = 1 ‘ Este es como la calidad media, no muy buena, pero tan poco ni muy mala.

Private Const NONANTIALIASED_QUALITY = 3 ‘ Se presentán sin antidentado

Private Const PROOF_QUALITY = 2

PitchAndFamily = Este parámetro debe ser la combinación de 2 constante. La primera determina el “Pitch” este puede tomar los siguientes valores:

Private Const FIXED_PITCH = 1

Private Const VARIABLE_PITCH = 2

Private Const DEFAULT_PITCH = 0

La segunda constante determina la familia a la cual pertenece la fuente, cuyos valores de la constante fueron especificados arriba cuando se hablo del tema. Para combinarlos hay que usar el operador “OR”. Ejm: FF_ROMAN Or VARIABLE_PITCH. En realidad el termino “Pitch” no lo tengo muy claro, por lo que dejo a las personas que quieran averiguar, leer la documentación oficial.

FaceName = Es una cadena que identifica el nombre de la fuente.

Como pueden ver cada parámetro tiene muchas opciones, y se pueden hacer muchas combinaciones, en nuestro proyecto trabajamos de manera sencilla:

fuente = CreateFont(-MulDiv(30, GetDeviceCaps(Me.hdc, LOGPIXELSY), 72), 0, 0, 0, 0, 0, 0, 0, DEFAULT_CHARSET, OUT_DEFAULT_PRECIS, CLIP_DEFUALT_PRECIS, ANTIALIASED_QUALITY, DEFAULT_PITCH, letra)

Ya que utilice en su mayor parte los valores por defecto. Excepto en ANTIALIASED_QUALITY en donde verán que las fuentes TrueType se ven de manera nítida, si quieren ver la diferencia quiten esta constante y pongan DEFAULT_QUALITY y notaran cierta diferencia.

La variable “fuente” es quien contiene nuestro nuevo objeto “fuente”, en caso de algún error en la creación de la fuente, la variable “fuente” contendrá NULL. En caso de ser diferente de NULL implica que la creación de la fuente fue exitosa. Y procedemos a colocarla en nuestro HDC.

    SelectObject Me.hdc, fuente

Y al final usamos:

     DeleteObject fuente

Para liberar el espacio de memoria.

Ahora que tenemos una nueva fuente, procedemos a usarla, para ello debemos de pintar un texto con las herramientas que Windows posee, y como dije trabajare solo con DrawText, la función mas sencilla es TextOut, pero como es sencilla les dejo a ustedes dicha documentación, si logran entender DrawText les aseguro que la función TextOut no la van a usar.

DrawText

Api: Declare Function DrawText Lib "user32" Alias "DrawTextA" (ByVal hdc As Long, ByVal lpStr As String, ByVal nCount As Long, lpRect As RECT, ByVal wFormat As Long) As Long

Esta función se encarga de pintar una cadena de texto en un HDC, el cual es su primer parámetro, como segundo parámetro tenemos la cadena de texto que queremos mostrar, como tercer parámetro tenemos la longitud de dicha cadena, como cuarto parámetro tenemos un rectángulo que delimita a dicha cadena. Y el parámetro más importante es “wFormat” que es la forma de cómo manipulamos la cadena de texto al momento de imprimirla.

Pueden ver que hay un rectángulo, eso significa que la cadena de texto se imprimirá en el recuadro que pasamos en “lpRect”, si la cadena es muy larga, ella no se imprime completa, a menos que sepamos manipular los valores que puede tener “wFormat”.

wFormat = puede tomar la combinación de los siguientes parámetro, antes de ir a estos valores noten que el proyecto tiene 7 bloques de comentarios y esta activado el último que es el mas sencillo, a medida que explique algunos valores de “wFormat” vamos a ir quitando los comentarios a algunas líneas.

Valores que puede tomar.

Private Const DT_BOTTOM = &H8

Utilizada par alinear el texto en la parte inferior del rectángulo. Es necesario combinarlo con DT_SINGLELINE

Private Const DT_CALCRECT = &H400

Este parámetro hace que no se imprima nada, lo único que hace es calcular el rectángulo que puede contener al texto, es decir, aquí la variable del parámetro “lpRect” se le pasa sin valores, la función lo que hace es calcular el ancho y alto que ocupa dicha cadena.

Private Const DT_CENTER = &H1

Utilizado para centrar el texto horizontalmente en el rectángulo especificado

Private Const DT_TOP = &H0

Alinea el texto en la parte superior del rectángulo

Private Const DT_VCENTER = &H4

Alinea el texto en el centro, pero verticalmente.

Private Const DT_LEFT = &H0

Alinea el texto hacia la izquierda del rectángulo

Private Const DT_EDITCONTROL = &H2000

Si se activa esta propiedad, le decimos a la función, que si el Texto no cabe en el rectángulo, que no lo imprima.

Private Const DT_END_ELLIPSIS = &H8000

Esta propiedad, hace que si la línea es muy grande y no cabe en el ancho del rectángulo, pues la función podrá cortar dicha cadena y ponerle 3 puntos al final, Ejm = “Hola Mun…”. Este parámetro solo es usado cuando la propiedad DT_SINGLELINE esta activada.

Private Const DT_EXTERNALLEADING = &H200

Este parámetro hace que se le sume a la altura de la línea la cantidad establecida por el diseñador de la fuente como distancia entre líneas.

Private Const DT_HIDEPREFIX = &H100000

Solo funcional en Windows 2000, lo que hace es que si existe un & el lo sustituye con un subrayado al carácter delante de él. Ejm “La Cade&na” = “La Cadena”, en caso de poner && entonces el sistema tomara el carácter &.

Private Const DT_NOPREFIX = &H800

Deshabilita la propiedad antes mencionada (DT_HIDEPREFIX).

Private Const DT_PREFIXONLY = &H200000

Igual que DT_HIDEPREFIX con la diferencia que no imprime el carácter. Ejm “La Cade&na” = “La Cade_a”

Private Const DT_INTERNAL = &H1000

Supongo que esta funciona si la usamos con el DT_CALCRECT, ya que lo que hace es realizar cualquier operación de cálculo, con la fuente del sistema, sin importar la fuente activada en el HDC

Private Const DT_NOCLIP = &H100

No toma en cuenta el rectángulo especificado en el parámetro “lpRect”

Private Const DT_RTLREADING = &H20000

En este caso la función imprimirá de derecha a izquierda, y no de izquierda a derecha. He de suponer que esto se utiliza en caso que deseen usar letras Árabes.

Private Const DT_SINGLELINE = &H20

Con esta propiedad lo que hacemos es indicarle a la función que solo imprima en una sola línea.

Private Const DT_WORDBREAK = &H10

Este es el caso Multiline, si el texto no cabe en una línea se imprime en las líneas que sean necesarias, por supuesto, verán lo que entre en el rectángulo.

En la documentación oficial encontraran un poco mas de constante, pero considero que hable la mas importantes.

NOTA: El rectángulo enmarcado que imprimí con la función Rectangle, no es mas que un dibujo indicativo, es decir, ese rectángulo visualiza el área de impresión del texto, siendo mas fácil entender las distintas constantes explicadas arriba.

Veamos el ejemplo más simple, que se encuentra en el proyecto:

    cadena = "Hola Mundo!"

    DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP

Imprimimos una cadena, alineada en la parte superior del rectángulo. Pero ahora veamos que pasa si jugamos con los parámetros. Pongan en comentario al bloque de código correspondiente al comentario 7 y quitenle los comentarios a las líneas del comentario 3:

    cadena = "Hola! esta es una cadena muyyyy larga"

    DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP Or DT_CENTER Or DT_WORDBREAK

Se darán cuenta que al pasar por diferentes tipos de fuentes, si la cadena no entra en el rectángulo, la cadena es cortada e impresa en otra línea. Pongan en comentario las líneas del comentario 3 y activen el comentario 4.

    cadena = "Hola! esta es una cadena muyyyy larga"

    DrawText Me.hdc, cadena, Len(cadena), r, DT_TOP Or DT_CENTER Or DT_END_ELLIPSIS Or DT_MODIFYSTRING

En este caso, si la línea no entra en el rectángulo, es cortada por la función mostrando 3 puntos al final.

Los demás comentarios serán explicados con las funciones nuevas que las acompaña. Y empecemos con el comentario 1. Si activamos las líneas del bloque comentario 1, verán que al imprimir se muestra un fondo azul, y las letras blancas.

SetBkMode

Api: Declare Function SetBkMode Lib "gdi32" (ByVal hdc As Long, ByVal nBkMode As Long) As Long

Con esta función establecemos si el fondo de los caracteres es transparente u opaco. Es decir, el parámetro “nBkMode” puede tomar 2 valores:

Const OPAQUE = 2

Const TRANSPARENT = 1

Es idéntico a lo que hace la propiedad “BackStyle” de los controles “Label”.

SetBkColor

Api: Declare Function SetBkColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long

En caso que el fondo sea Opaco, se tomara el color que se le pase a esta función en su segundo parámetro “crColor”, en caso de ser transparente, pues no se toma en cuenta.

SetTextColor

Api: Declare Function SetTextColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long

Con esta función determinamos el color de las letras. El color se le pasa en el segundo parámetro.

GetTextExtentPoint32

Api: Declare Function GetTextExtentPoint32 Lib "gdi32" Alias "GetTextExtentPoint32A" (ByVal hdc As Long, ByVal lpsz As String, ByVal cbString As Long, lpSize As Size) As Long

Esta función sirve para calcular la longitud en unidades lógicas de una cadena. En el parámetro “lpsz” se le pasa dicha cadena,  en el parámetro “cbString” se le pasa la longitud de la cadena, y en el parámetro “lpSize”, se la pasa una variable de tipo Size, que contiene un X e Y. X es lo que abarca la cadena Horizontalmente e Y tendrá la longitud vertical.

En el ejemplo, se usa la función Rectangle para enmarcar dicha cadena. Para verlo en el ejemplo quiten los comentarios al bloque de código “comentario 2”.

SetTextJustification

Api: Declare Function SetTextJustification Lib "gdi32" (ByVal hdc As Long, ByVal nBreakExtra As Long, ByVal nBreakCount As Long) As Long

Esta función permite alinear un texto de manera Justificada. El problema esta en los parámetros que hay que pasarles a la función para que ella haga la alineación. Primero le pasamos el hdc, el segundo parámetro es el espacio vacío, es decir:




En el tercer parámetro le indicamos cuantos espacios en blanco hay entre cada palabra. En nuestro ejemplo usamos "Hola otra vez Mundo!", como pueden ver esta cadena posee 4 espacios en blanco.

Con estos datos la función adapta la cadena según el ancho especificado en nBreakExtra.

Para ver el ejemplo de esta función quitenle los comentarios al bloque de líneas “comentario 5”.

SetTextCharacterExtra

Api: Declare Function SetTextCharacterExtra Lib "gdi32" (ByVal hdc As Long, ByVal nCharExtra As Long) As Long

Un  carácter tiene cierta separación de otro carácter, esa separación puede ser controlada, esta es la función a cargo de controlar ese espacio. Es fácil ver lo que hace, quitenle los comentarios al bloque del comentario 6. Este espacio es pasado a través del parámetro “nCharExtra”.

Otras funciones

En esta sección trabajamos algunas funciones que no se usan en el proyecto, pero son importantes mencionarlas.

GetTextCharacterExtra

Api: Declare Function GetTextCharacterExtra Lib "gdi32" (ByVal hdc As Long) As Long

GetTextColor

Api: Declare Function GetTextColor Lib "gdi32" (ByVal hdc As Long) As Long

GetBkColor

Api: Declare Function GetBkColor Lib "gdi32" (ByVal hdc As Long) As Long

GetBkMode

Api: Declare Function GetBkMode Lib "gdi32" (ByVal hdc As Long) As Long

Normalmente toda función que empieza con Set tiene su Get, con estas 5 funciones lo que hacemos es tomar el valor correspondiente dependiendo de la función. Basta con asignar a una variable el valor que retorna la función. 

GetTextMetrics

Api: Declare Function GetTextMetrics Lib "gdi32" Alias "GetTextMetricsA" (ByVal hdc As Long, lpMetrics As TEXTMETRIC) As Long

Esta función es utilizada para tomar la información correspondiente a la fuente que se encuentra en el dispositivo de contexto. Esa información se encuentra en la variables “lpMetricS” que es de tipo TEXTMETRIC, parámetros explicados con anterioridad.

MulDiv

Api: Declare Function MulDiv Lib "kernel32" (ByVal nNumber As Long, ByVal nNumerator As Long, ByVal nDenominator As Long) As Long

Esta función multiplica 2 valores de 32 Bits, y el resultado de 64 Bits lo divide por un tercer valor de 32 Bits.

nNumber = es el multiplicando.

nNumerator = es con el que se multiplica.

nDenominator =  es el denominador de dividir la multiplicación de los dos de arriba.

Si la función es exitosa ella retorna dicho resultado, de lo contrario retorna –1.

La función GetDeviceCaps la explicare en un Capítulo  donde ella se encuentra.

Capítulo  IV.  BITMAP (MAPA DE BITS)

A continuación se presentaran temas relacionados con la manipulación de BITMAP (Mapa de bits), las cuales pueden ser de mucha ayuda al desarrollar una aplicación que las contengan. 

Cabe mencionar que las medidas que se usan en estas funciones son en unidades logicas, es decir, parámetros como X, Y, Width y Height, vienen especificado en unidades logicas.

COPIANDO IMAGENES

Para este proyecto se necesita lo siguiente:

20 PicturesBox

10 Command o botones,

Tienen que colocar los PictureBox de esta manera, el picture1 va al lado del picture2, el picture3 al lado del picture4, y así sucesivamente, el tamaño de los PictureBox en píxel es 60X60, es decir, tendrán una columna de impares y una columnas de pares, luego al lado del par picture1 y picture2 colocan Command1, de picture3 y picture4 colocan Command2 y así sucesivamente, y a los PictureBox impares le colocan la imagen azul.jpg, que viene con este documento, y a los PictureBox pares le colocan rojo.jpg.

Y luego pegan este código:

Private Declare Function BitBlt Lib "gdi32" (ByVal hDestDC As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal dwRop As Long) As Long

Private Const BLACKNESS = &H42

Private Const DSTINVERT = &H550009

Private Const MERGECOPY = &HC000CA

Private Const MERGEPAINT = &HBB0226

Private Const NOTSRCCOPY = &H330008

Private Const NOTSRCERASE = &H1100A6

Private Const PATCOPY = &HF00021

Private Const PATINVERT = &H5A0049

Private Const PATPAINT = &HFB0A09

Private Const SRCAND = &H8800C6

Private Const SRCCOPY = &HCC0020

Private Const SRCERASE = &H440328

Private Const SRCINVERT = &H660046

Private Const SRCPAINT = &HEE0086

Private Const WHITENESS = &HFF0062

Private Sub Command1_Click()

    BitBlt Picture2.hdc, 0, 0, Picture1.Width, Picture1.Height, Picture1.hdc, 0, 0, BLACKNESS

End Sub

Private Sub Command2_Click()

    BitBlt Picture4.hdc, 0, 0, Picture4.Width, Picture4.Height, Picture3.hdc, 0, 0, WHITENESS

End Sub

Private Sub Command3_Click()

    BitBlt Picture6.hdc, 0, 0, Picture6.Width, Picture6.Height, Picture5.hdc, 0, 0, SRCCOPY

End Sub

Private Sub Command4_Click()

    BitBlt Picture8.hdc, 0, 0, Picture8.Width, Picture8.Height, Picture7.hdc, 0, 0, SRCAND

End Sub

Private Sub Command5_Click()

    BitBlt Picture10.hdc, 0, 0, Picture10.Width, Picture10.Height, Picture9.hdc, 0, 0, SRCPAINT

End Sub

Private Sub Command6_Click()

    BitBlt Picture12.hdc, 0, 0, Picture12.Width, Picture12.Height, Picture11.hdc, 0, 0, SRCERASE

End Sub

Private Sub Command7_Click()

    BitBlt Picture14.hdc, 0, 0, Picture14.Width, Picture14.Height, Picture13.hdc, 0, 0, SRCINVERT

End Sub

Private Sub Command8_Click()

    BitBlt Picture16.hdc, 0, 0, Picture16.Width, Picture16.Height, Picture15.hdc, 0, 0, NOTSRCCOPY

End Sub

Private Sub Command9_Click()

    BitBlt Picture18.hdc, 0, 0, Picture18.Width, Picture18.Height, Picture17.hdc, 0, 0, NOTSRCERASE

End Sub

Private Sub Command10_Click()

    BitBlt Picture20.hdc, 0, 0, Picture20.Width, Picture20.Height, Picture19.hdc, 0, 0, DSTINVERT

End Sub

Arranquen el programa y presionen a los botones y se darán cuenta de lo que sucede, ustedes dirán ¿Eduardo que esto?. En esta sección introduzco una nueva API que es BitBlt. Esta API en pocas palabras lo que hace es copiar una imagen fuente a un dispositivo destino. Pero ustedes dirán: estas mintiendo Eduardo ya que algunos de los cuadritos aparecen con otro color,  y yo les digo: lo que ocurre es que hay diferentes formas de copiar una imagen aplicando ciertos cambios lógicos. Como lo veremos a continuación:

BitBlt

Api: Declare Function BitBlt Lib "gdi32" (ByVal hDestDC As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal dwRop As Long) As Long

Ya dije que esta API se usa para copiar un bitmap de un lugar a otro, es decir, desde un dispositivo de contexto fuente a un dispositivo de contexto destino, analicemos sus parámetros:

hDestDC: dispositivo de contexto DESTINO, es decir, quien recibe la imagen.

x, y: estas dos variables determinan el vértice izquierdo superior donde vamos a  copiar.

nWidth, nHeight: determinan el ancho y alto de la copia.

hSrcDC: dispositivo de contexto de ORIGEN, es decir, de donde se saca la imagen a copiar.

xSrc, ySrc: estas dos variables determinan el vértice izquierdo superior donde vamos a empezar a copiar.

dwRop: Constante que determinan el tipo de copia que se va hacer. Esta variable puede tomar los siguientes valores:

Const BLACKNESS = &H42

Const DSTINVERT = &H550009

Const MERGECOPY = &HC000CA

Const MERGEPAINT = &HBB0226

Const NOTSRCCOPY = &H330008

Const NOTSRCERASE = &H1100A6

Const PATCOPY = &HF00021

Const PATINVERT = &H5A0049

Const PATPAINT = &HFB0A09

Const SRCAND = &H8800C6

Const SRCCOPY = &HCC0020

Const SRCERASE = &H440328

Const SRCINVERT = &H660046

Const SRCPAINT = &HEE0086

Const WHITENESS = &HFF0062

Creo que los parámetros son fáciles de entender hasta dwRop, para explicar este parámetro vamos al código:

Empecemos con command1:

    BitBlt Picture2.hdc, 0, 0, Picture1.Width, Picture1.Height, Picture1.hdc, 0, 0, BLACKNESS

Vamos a leer en castellano esta línea de código: usamos BitBlt cuyo HDC destino es PICTURE2.HDC, lo que recibas lo copias empezando por el vértice 0,0, el tamaño de la copia será Picture1.width, Picture1.height, y se copiara la imagen que esta en Picture1.hdc, y empiezas a copiar desde el punto 0,0, y vas a usar BLACKNESS para copiar.

El “castellano” se entiende hasta BLACKNEES. A continuación describimos que hace cada constante:

Const BLACKNESS = &H42

Rellena el destino, con el color asociado al primer índice de la paleta. En este caso como no hemos creado ninguna paleta, pues usa la paleta por defecto cuyo primer elemento (0) es el negro, por esa razón vemos negro en el PictureBox destino. Si tuviéramos una paleta de colores asociada, pues entonces tomaría el color perteneciente al índice 0 que no necesariamente tendría que ser negro.

Const WHITENESS = &HFF0062

Lo mismo que el anterior pero toma el segundo elemento, es decir, el índice (1), que en la paleta por defecto es el blanco.

Const SRCCOPY = &HCC0020

Hace una copia exacta desde el origen hasta el destino, por eso ven el color azul en el PictureBox destino.

Const SRCAND = &H8800C6

Hace una copia usando el AND lógico para establecer los píxeles de la imagen destino, y ustedes dirán ¿Qué significa eso? ¿Por qué veo negro en el resultado?.

Pues fácil: el color azul se escribe de esta manera en Hexadecimal 0000FF, Los primeros 00 corresponden al ROJO los segundo 00 corresponden al verde y el FF corresponden al Azul, es decimal sería 0 = rojo, 0 = verde, y 255 = azul, es fácil ver esto si lo ven con el PAINT de Windows o cualquier programa grafico, ya que estas imágenes son de 24 bits, significa que hay 3 bytes por color, que sería el RGB, es decir, el rojo, verde y azul equivalen a un 1 byte c/u = 8 bits , y el valor máximo que podemos alcanzar con 8 bits es 255.

Lo que nos interesa es 0000FF que en binario sería algo así:

ROJO = 00000000 = 0

VERDE = 00000000 = 0 

AZUL = 11111111 = 255 

Es decir, RGB(0,0,255) o 0000FF

Estamos hablando del PictureBox origen que es de color azul, el otro picture destino tiene color rojo que sería:

ROJO = 11111111= 255

VERDE = 00000000 = 0 

AZUL = 00000000 = 0 

Es decir, RGB(255,0,0) o FF0000

Pongamos esto en dos líneas:

Color del PictureBox origen: 
00000000
00000000
11111111

Color del PictureBox destino:
11111111
00000000
00000000

Si aplicamos un AND lógico =
00000000
00000000
00000000

Es decir, el resultado es RGB(0,0,0) ¿Qué es este color? Pues negro, por esa razón es que el resultado da negro ¿Claro? Si no esta claro revisen sus apuntes de lógica, que algunas ves debieron ver en la carrera de informática.

Const PATPAINT = &HFB0A09

Equivale al operador lógico O. El resultado sería así:

Color del PictureBox origen: 
00000000
00000000
11111111

Color del PictureBox destino:
11111111
00000000
00000000

Si aplicamos un O lógico =
11111111
00000000
11111111

Y tenemos RGB(255,0,255). Lo que equivale al color que ven como resultado, si no me creen, abran paint, colóquense en la sección de editar colores y coloquen R = 255, G = 0 y B = 255 y verán el mismo color.

Const SRCERASE = &H440328

Este lo que hace es: Invierte los colores del PictureBox o HDC destino y luego aplica el operador AND con el PictureBox fuente:

Color del PictureBox destino:
11111111
00000000
00000000

Invertimos


00000000
11111111
11111111

Color del PictureBox origen: 
00000000
00000000
11111111

Color Invertido destino

00000000
11111111
11111111

Si aplicamos un AND lógico =
00000000
00000000
11111111

Y tenemos el azul de nuevo.

Const SRCINVERT = &H660046

Aplica el operador lógico XOR.

Color del PictureBox origen: 
00000000
00000000
11111111

Color del PictureBox destino:
11111111
00000000
00000000

Si aplicamos un XOR  lógico =
11111111
00000000
11111111

Pero vamos aplicar de nuevo un XOR, pero con la imágen cambiada, es decir:

Color del PictureBox origen: 
00000000
00000000
11111111

Color del PictureBox destino:
11111111
00000000
11111111 

Si aplicamos un XOR  lógico =
11111111
00000000
00000000

Pueden ver que el resultado es la imagen que estaba antes, por esa razón, si ustedes aprietan 2 veces el botón, verán primero el violeta que es el resultado del primer XOR, y luego verán el color que estaba originalmente, este comando o constante es muy usado cuando quieres copiar imágenes y deshacer el cambio.

Const NOTSRCCOPY = &H330008

Invierte los colores del dispositivo de origen y lo copia en el destino, sin importar lo que tenga el destino:

Color del PictureBox origen: 
00000000
00000000
11111111

Invertimos


11111111
11111111
00000000

Const NOTSRCERASE = &H1100A6

Combina los colores usando O y luego los invierte:

Color del PictureBox origen: 
00000000
00000000
11111111

Color del PictureBox destino:
11111111
00000000
00000000

Si aplicamos un O lógico =
11111111
00000000
11111111

Invertimos el resultado

00000000
11111111
00000000

Y tenemos VERDE.

Const DSTINVERT = &H550009

Invierte los colores del destino sin importar la imagen fuente.

Color del PictureBox destino:
11111111
00000000
00000000

Invierte



00000000
11111111
11111111

Y tenemos Agua Marine o Azul Claro.

Const PATCOPY = &HF00021

Const PATINVERT = &H5A0049

Const PATPAINT = &HFB0A09

Estas constantes hacen lo mismo que sus equivalentes, SRCCOPY, SCRINVERT y SRCPAINT, respectivamente, pero con las brochas. 

El hecho que haya seleccionado imágenes UNICOLORES fue para que vieran el efecto rápidamente, pero ese cálculo lógico se hace con cualquier imagen de cualquier color. 

ENCOGIENDO Y ESTIRANDO IMAGENES

Para este proyecto necesitamos 8 pictureBox 4 botones (dejen los nombres que VB pone por defecto), a los PictureBox impares (Picture1, Picture3, etc), coloquenle la imagen “imagen4.jpg” que acompaña a este proyecto. Luego peguen este código:

Private Declare Function StretchBlt Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal nSrcWidth As Long, ByVal nSrcHeight As Long, ByVal dwRop As Long) As Long

Private Const SRCCOPY = &HCC0020

Private Sub Form_Load()

    Picture1.ScaleMode = 3

    Picture1.Height = 1455

    Picture1.Width = 1500

    Picture1.Top = 405

    Picture1.Left = 405

    Picture3.ScaleMode = 3

    Picture3.Height = 1455

    Picture3.Width = 1500

    Picture3.Top = 1980

    Picture3.Left = 405

    Picture5.ScaleMode = 3

    Picture5.Height = 1455

    Picture5.Width = 1500

    Picture5.Top = 3960

    Picture5.Left = 405

    Picture7.ScaleMode = 3

    Picture7.Height = 1455

    Picture7.Width = 1500

    Picture7.Top = 5670

    Picture7.Left = 405

    Picture2.ScaleMode = 3

    Picture2.Height = 915

    Picture2.Width = 870

    Picture2.Top = 405

    Picture2.Left = 2070

    Picture4.ScaleMode = 3

    Picture4.Height = 1680

    Picture4.Width = 1995

    Picture4.Top = 1980

    Picture4.Left = 2070

    Picture6.ScaleMode = 3

    Picture6.Height = 1455

    Picture6.Width = 1500

    Picture6.Top = 3960

    Picture6.Left = 2115

    Picture8.ScaleMode = 3

    Picture8.Height = 1455

    Picture8.Width = 1500

    Picture8.Top = 5670

    Picture8.Left = 2115

    Command1.Top = 630

    Command1.Left = 3195

    Command1.Width = 1230

    Command1.Height = 465

    Command1.Caption = "disminuir"

    Command2.Top = 2610

    Command2.Left = 4230

    Command2.Width = 1230

    Command2.Height = 465

    Command2.Caption = "aumentar"

    Command3.Top = 4455

    Command3.Left = 3825

    Command3.Width = 1230

    Command3.Height = 465

    Command3.Caption = "espejo vertical"

    Command4.Top = 6120

    Command4.Left = 3825

    Command4.Width = 1230

    Command4.Height = 465

    Command4.Caption = "espejo horizontal"

    Me.Height = 7710

    Me.Width = 5865

End Sub

Private Sub Command1_Click()

    StretchBlt Picture2.hdc, 0, 0, Picture2.ScaleWidth, Picture2.ScaleHeight, Picture1.hdc, 0, 0, Picture1.ScaleWidth, Picture1.ScaleHeight, SRCCOPY

End Sub

Private Sub Command2_Click()

    StretchBlt Picture4.hdc, 0, 0, Picture4.ScaleWidth, Picture4.ScaleHeight, Picture3.hdc, 0, 0, Picture3.ScaleWidth, Picture3.ScaleHeight, SRCCOPY

End Sub

Private Sub Command3_Click()

    StretchBlt Picture6.hdc, 0, 0, Picture6.ScaleWidth, Picture6.ScaleHeight, Picture5.hdc, 0, Picture5.ScaleHeight, Picture5.ScaleWidth, -Picture5.ScaleHeight, SRCCOPY

End Sub

Private Sub Command4_Click()

    StretchBlt Picture8.hdc, 0, 0, Picture8.ScaleWidth, Picture8.ScaleHeight, Picture7.hdc, Picture7.ScaleWidth, 0, -Picture7.ScaleWidth, Picture7.ScaleHeight, SRCCOPY

End Sub

Es sencillo ver lo que hace el programa al presionar los distintos botones. Pues solo trabajamos con una sola función API (StretchBlt), que explicamos a continuación:

StretchBlt

Api: Declare Function StretchBlt Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal nSrcWidth As Long, ByVal nSrcHeight As Long, ByVal dwRop As Long) As Long

Esta función se encarga de encoger o alargar una imagen, y manipulando algunos parámetros podemos obtener un efecto de espejo.

Parámetros:

hdc = dispositivo de contexto “destino”, es decir, donde ira la imagen transformada

x, y, nWidth, nHeight = es como el “recuadro” destino o donde se ubicara la imagen seleccionada.

hSrcDC = dispositivo de contexto que contiene la imagen a transformar.

xSrc, ySrc, nSrcWidth, nSrcHeight = es como el “recuadro” origen o donde se encuentra la imagen a manipular.

OJO: en el proyecto ustedes ven, como la imagen en los primeros 2 botones, se encoge y se estira, pero ustedes pueden tomar, pasando bien los parámetros de arriba, porciones de la imagen y estirar o encoger esa porción, es decir, no tienen por que trabajar con la imagen completa.

dwRop = toma exactamente las mismas constantes que la función Bitblt. Y funciona exactamente igual que Bitblt.

Vean que para aplicar los efectos de espejo, lo único que hay que hacer es poner en negativo algunos valores.

DEGRADANDO COLORES (GRADIENTE)

Esta función es muy bonita, pero no funciona en Windows 95, es decir, sirve para sistemas operativos del 98 en adelante. Para este proyecto solo tienen que pegar este código al formulario:

Option Explicit

Private Type TRIVERTEX

    X As Long

    Y As Long

    Red As Integer

    Green As Integer

    Blue As Integer

    Alpha As Integer

End Type

Private Type GRADIENT_RECT

    UpperLeft As Long

    LowerRight As Long

End Type

Private Type GRADIENT_TRIANGLE

    Vertex1 As Long

    Vertex2 As Long

    Vertex3 As Long

End Type

Const GRADIENT_FILL_RECT_H As Long = &H0

Const GRADIENT_FILL_RECT_V  As Long = &H1

Const GRADIENT_FILL_TRIANGLE As Long = &H2

Const GRADIENT_FILL_OP_FLAG As Long = &HFF

Private Declare Function GradientFillRect Lib "msimg32" Alias "GradientFill" (ByVal hDC As Long, pVertex As TRIVERTEX, ByVal dwNumVertex As Long, pMesh As GRADIENT_RECT, ByVal dwNumMesh As Long, ByVal dwMode As Long) As Long

Private Declare Function GradientFillTriangle Lib "msimg32" Alias "GradientFill" (ByVal hDC As Long, pVertex As TRIVERTEX, ByVal dwNumVertex As Long, pMesh As GRADIENT_TRIANGLE, ByVal dwNumMesh As Long, ByVal dwMode As Long) As Long

Dim m_ColorIzqR As Integer

Dim m_ColorIzqG As Integer

Dim m_ColorIzqB As Integer

Dim m_ColorDerR As Integer

Dim m_ColorDerG As Integer

Dim m_ColorDerB As Integer

Dim colortriangulo(0 To 4) As TRIVERTEX

Dim triangulos(0 To 3) As GRADIENT_TRIANGLE

Dim tipo As Boolean

Dim resultado As Integer

Private Sub Form_Load()

    Me.WindowState = 2

    resultado = MsgBox("Desea ver el degradado de manera triangular o rectangular, si lo desea en forma triangular presiona SI en caso de quererlo en forma rectangular presione NO", vbYesNo, "")

    If resultado = vbNo Then

        CommonDialog1.DialogTitle = "Color del Lado Izquierdo"

        CommonDialog1.ShowColor

        m_ColorIzqR = ColorRojo(CommonDialog1.Color)

        m_ColorIzqG = ColorVerde(CommonDialog1.Color)

        m_ColorIzqB = ColorAzul(CommonDialog1.Color)

        CommonDialog1.DialogTitle = "Color del Lado Derecho"

        CommonDialog1.ShowColor

        m_ColorDerR = ColorRojo(CommonDialog1.Color)

        m_ColorDerG = ColorVerde(CommonDialog1.Color)

        m_ColorDerB = ColorAzul(CommonDialog1.Color)

        tipo = True

    End If

        Me.ScaleMode = vbPixels

End Sub

Private Sub Form_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)

If Button = 1 Then

    CommonDialog1.DialogTitle = "Color del Lado Izquierdo"

    CommonDialog1.ShowColor

    m_ColorIzqR = ColorRojo(CommonDialog1.Color)

    m_ColorIzqG = ColorVerde(CommonDialog1.Color)

    m_ColorIzqB = ColorAzul(CommonDialog1.Color)

    Me.Refresh

End If

If Button = 2 Then

    CommonDialog1.DialogTitle = "Color del Lado Derecho"

    CommonDialog1.ShowColor

    m_ColorDerR = ColorRojo(CommonDialog1.Color)

    m_ColorDerG = ColorVerde(CommonDialog1.Color)

    m_ColorDerB = ColorAzul(CommonDialog1.Color)

    Me.Refresh

End If

End Sub

Private Sub Form_Paint()

If tipo Then

    Dim vert(1) As TRIVERTEX

    Dim gRect As GRADIENT_RECT

    With vert(0)

        .X = 0

        .Y = 0

        .Red = m_ColorIzqR

        .Green = m_ColorIzqG

        .Blue = m_ColorIzqB

        .Alpha = 0&

    End With

    With vert(1)

        .X = Me.ScaleWidth

        .Y = Me.ScaleHeight

        .Red = m_ColorDerR

        .Green = m_ColorDerG

        .Blue = m_ColorDerB

        .Alpha = 0&

    End With

    gRect.UpperLeft = 0

    gRect.LowerRight = 1

    GradientFillRect Me.hDC, vert(0), 2, gRect, 1, GRADIENT_FILL_RECT_H

Else

    colortriangulo(0).X = 0

    colortriangulo(0).Y = 0

    colortriangulo(0).Alpha = 0

    colortriangulo(0).Blue = ColorAzul(&HFF0000)

    colortriangulo(0).Red = ColorRojo(&HFF0000)

    colortriangulo(0).Green = ColorVerde(&HFF0000)

    colortriangulo(1).X = Me.ScaleWidth

    colortriangulo(1).Y = 0

    colortriangulo(1).Alpha = 0

    colortriangulo(1).Blue = ColorAzul(&HFFFF&)

    colortriangulo(1).Red = ColorRojo(&HFFFF&)

    colortriangulo(1).Green = ColorVerde(&HFFFF&)

    colortriangulo(2).X = Me.ScaleWidth / 2

    colortriangulo(2).Y = Me.ScaleHeight / 2

    colortriangulo(2).Alpha = 0

    colortriangulo(2).Blue = ColorAzul(&HFFFFFF)

    colortriangulo(2).Red = ColorRojo(&HFFFFFF)

    colortriangulo(2).Green = ColorVerde(&HFFFFFF)

    colortriangulo(3).X = 0

    colortriangulo(3).Y = Me.ScaleHeight

    colortriangulo(3).Alpha = 0

    colortriangulo(3).Blue = ColorAzul(&HFF&)

    colortriangulo(3).Red = ColorRojo(&HFF&)

    colortriangulo(3).Green = ColorVerde(&HFF&)

    colortriangulo(4).X = Me.ScaleWidth

    colortriangulo(4).Y = Me.ScaleHeight

    colortriangulo(4).Alpha = 0

    colortriangulo(4).Blue = ColorAzul(&HC000C0)

    colortriangulo(4).Red = ColorRojo(&HC000C0)

    colortriangulo(4).Green = ColorVerde(&HC000C0)

    triangulos(0).Vertex1 = 0

    triangulos(0).Vertex2 = 2

    triangulos(0).Vertex3 = 1

    triangulos(1).Vertex1 = 0

    triangulos(1).Vertex2 = 3

    triangulos(1).Vertex3 = 2

    triangulos(2).Vertex1 = 3

    triangulos(2).Vertex2 = 2

    triangulos(2).Vertex3 = 4

    triangulos(3).Vertex1 = 4

    triangulos(3).Vertex2 = 2

    triangulos(3).Vertex3 = 1

    GradientFillTriangle Me.hDC, colortriangulo(0), 5, triangulos(0), 4, GRADIENT_FILL_TRIANGLE

End If

End Sub

Function ColorAzul(ByVal New_GradienteOrientacion As Long) As Integer

On Error GoTo Errores:

    Dim m_Color As Long

    m_Color = New_GradienteOrientacion

    Dim Blue As Long

    Dim Color As String

    Color = Hex(New_GradienteOrientacion)

    If Len(Color) = 2 Then

        Color = "0000" & Color

    Else

        If Len(Color) = 4 Then Color = "00" & Color

    End If

    If Val("&H" & Mid(Color, 1, 2)) = 0 Then

        Blue = 0

    Else

        Blue = (255& * Val("&H" & Mid(Color, 1, 2))) + Val("&H" & Mid(Color, 1, 2)) - 65536

    End If

    ColorAzul = Blue

Errores:

    If Err.Number = 6 Then

        Blue = 0

        Resume

    End If

End Function

Function ColorVerde(ByVal New_GradienteOrientacion As Long) As Integer

On Error GoTo Errores:

    Dim m_Color As Long

    m_Color = New_GradienteOrientacion

    Dim Green As Long

    Dim Color As String

    Color = Hex(New_GradienteOrientacion)

    If Len(Color) = 2 Then

        Color = "0000" & Color

    Else

        If Len(Color) = 4 Then Color = "00" & Color

    End If

    If Val("&H" & Mid(Color, 3, 2)) = 0 Then

        Green = 0

    Else

        Green = (255& * Val("&H" & Mid(Color, 3, 2))) + Val("&H" & Mid(Color, 3, 2)) - 65536

    End If

    ColorVerde = Green

Errores:

    If Err.Number = 6 Then

        Green = 0

        Resume

    End If

End Function

Function ColorRojo(ByVal New_GradienteOrientacion As Long) As Integer

On Error GoTo Errores:

    Dim m_Color As Long

    m_Color = New_GradienteOrientacion

    Dim Red As Long

    Dim Color As String

    Color = Hex(New_GradienteOrientacion)

    If Len(Color) = 2 Then

        Color = "0000" & Color

    Else

        If Len(Color) = 4 Then Color = "00" & Color

    End If

    If Val("&H" & Mid(Color, 5, 2)) = 0 Then

        Red = 0

    Else

        Red = (255& * Val("&H" & Mid(Color, 5, 2))) + Val("&H" & Mid(Color, 5, 2)) - 65536

    End If

    ColorRojo = Red

Errores:

    If Err.Number = 6 Then

        Red = 0

        Resume

    End If

End Function

Cuando arrancan el proyecto verán un Msgbox que le indica el como quieren el degradado. Hay dos maneras de hacerlo de manera triangular o rectangular, si presionan “NO” al Msgbox, el programa les pedirá 2 colores, traten de colocar colores bien distintos para que vean el efecto, luego verán que el formulario presenta una degradación de un color a otro. En caso de presionar “SI” el programa muestra un degradado triangular.

Para esas degradaciones existen 2 funciones dedicadas a ello. Una se encarga de la rectangular y la otra de la triangular. Presten atención,  ya que las funciones tienen ciertos detalles a tener en cuenta. Empecemos por la más sencilla que es la rectangular.

GradientFillRect

Api: Declare Function GradientFillRect Lib "msimg32" Alias "GradientFill" (ByVal hDC As Long, pVertex As TRIVERTEX, ByVal dwNumVertex As Long, pMesh As GRADIENT_RECT, ByVal dwNumMesh As Long, ByVal dwMode As Long) As Long

Esta funciones no son tan fáciles de entender, una de las primeras cosas a entender es que el hecho que tenga como parámetro un HDC, no implica que la degradación o gradiente se aplica a todo el HDC, se pueden tener 10 rectángulos ubicados en distintas partes del HDC y cada rectángulo poseer una degradación distinta de los demás. Así que presten mucha atención a los parámetros.

pVertex = es una estructura del tipo TRIVERTEX que se representa de la siguiente manera:

Private Type TRIVERTEX

    X As Long

    Y As Long

    Red As Integer

    Green As Integer

    Blue As Integer

    Alpha As Integer

End Type

Esta estructura configura las esquinas de un rectángulo, con cada color, luego Windows se encarga de hacer el gradiente o degradado de una esquina a otra. 

Hay un detalle importante vean que el color se tiene que separar en RED, GREEN, y BLUE esto nunca antes lo habían usado, siempre usábamos la función RGB para establecer un color, pero ahora la función RGB no funciona de mucho, ya que se necesitan los colores por separado. Es por esa razón que tengo 3 funciones: ColorAzul, ColorVerde, ColorRojo, estas funciones lo que hacen es que dado un color en formato RGB, descomponen cada una de sus componentes, por ejemplo ColorAzul, recibe el valor del color y esta función retorna solo el color azul.

Recuerden algo, los colores no son mas que números, en Internet para resolver este problema lo que hacían era usar un calculo hexadecimal, no recuerdo por que yo no use eso, lo que sucede es que yo quería que funcionara para cualquier color y creo que esa técnica no funcionaba, y como consecuencia tuve que diseñar mis propias funciones, pero aquellos que puedan encontrar una forma mas rápida y efectiva de separar los colores, pues usenlo.

El parámetro Alpha de la estructura según la documentación es para uso futuro.

dwNumVertex = determina el número de vértices. Que en nuestro caso son 2.

pMesh = pasamos un arreglo con el orden de los vértices.

dwNumMesh = determinamos cuantos rectángulos tenemos.

dwMode = determina como queremos el gradiente, es decir si vertical (arriba-abajo) u horizontal (izquierda-derecha). Puede tomar dos valores:

Const GRADIENT_FILL_RECT_H As Long = &H0 ‘ Horizontal
Const GRADIENT_FILL_RECT_V  As Long = &H1 ‘ Vertical
Para verlo mas claro analicemos el ejemplo:

vert(0), es nuestro primer vértice el cual empezamos en 0,0 y le configuramos sus respectivos RGB por separados asignándole las variables que diseñe para ellos. El llenado de esas variables lo ven en el evento Form_Load. Luego vert(1) determina la esquina inferior derecha, la cual toma los valores de altura y anchura del respectivo formulario.

Llenamos la variable “gRect” de esta manera:

    gRect.UpperLeft = 0

    gRect.LowerRight = 1
Este 0 en “UpperLeft”, significa que la esquina superior izquierda se corresponde al elemento 0 del vector o arreglo “vert”, y 1 significa lo mismo pero para la posición número 1 del mismo vector.

A continuación llamamos a la función:

GradientFillRect Me.hDC, vert(0), 2, gRect, 1, GRADIENT_FILL_RECT_H

Aquí le decimos a Windows lo siguiente: “en el HDC = Me.hDC, dibuja un gradiente cuyo vértices empiezan en el vector “vert(0)” a su vez le indicamos que son 2, luego te paso el orden de cómo configuro los vértices, en la variable “gRect”, le indicamos que queremos 1 rectángulo, y al final le indicamos que el gradiente lo haga de manera horizontal”.

Ahora vamos a una función un poquito compleja ya que es triangular, se trabaja de la misma manera pero en ves de 2 vértices tendremos 3 y yo lo complique un poquito al no graficar solo un triangulo sino 4.

GradientFillTriangle

Api: Declare Function GradientFillTriangle Lib "msimg32" Alias "GradientFill" (ByVal hDC As Long, pVertex As TRIVERTEX, ByVal dwNumVertex As Long, pMesh As GRADIENT_TRIANGLE, ByVal dwNumMesh As Long, ByVal dwMode As Long) As Long

Los parámetros son exactamente iguales a su homologa de arriba, con la diferencia que al final al parámetro “dwMode” le ponemos GRADIENT_FILL_TRIANGLE. Indicando que va a rellenar el triangulo que configuramos.

Ustedes cuando presionan “SI” en el Msgbox ven 4 triángulos que se detallan un poquito en la figura de abajo:

[image: image10.jpg]



Las líneas de color crema son colocadas a propósito, para que distingan los 4 triángulos. Todos ustedes inmediatamente intuyen que se necesitan 12 vértices para los 4 triángulos, en realidad si se necesitan, pero como en nuestro ejemplo si se dan cuenta, nuestro centro es blanco, es decir, un vértice de cada triangulo tiene el color blanco, en nuestro caso es el vértice del centro. Por consiguiente, en nuestro ejemplo usamos solo 5 vértices y no 12.

¿Cómo hacemos eso?

Pueden ver que el vector o arreglo “coloretriangulo” tiene configurado 5 vértices, estos vértices se configuran igual que el rectángulo.

Lo importante esta en el vector “triángulos”, ya que este arreglo que a su ves contiene una estructura de tipo TRIVERTEX, configura como son cada triangulo. Analicemos el código:

Triangulo 1

triángulos(0).Vertex1 = 0 ‘ Este vértice toma la configuración puesta en colortriangulo (0)

triángulos(0).Vertex2 = 2 ‘ Este vértice toma la configuración puesta en colortriangulo (2)

triángulos(0).Vertex3 = 1 ‘ Este vértice toma la configuración puesta en colortriangulo (1)

Triangulo 2

triángulos(1).Vertex1 = 0 ‘ Este vértice toma la configuración puesta en colortriangulo (0)

triángulos(1).Vertex2 = 3 ‘ Este vértice toma la configuración puesta en colortriangulo (3)

triángulos(1).Vertex3 = 2 ‘ Este vértice toma la configuración puesta en colortriangulo (2)

Y así sucesivamente con los 2 restantes, dense cuenta que todos los triángulos tienen un vértice en común es el número 2. Que si vemos su configuración es la siguiente:

    colortriangulo(2).X = Me.ScaleWidth / 2

    colortriangulo(2).Y = Me.ScaleHeight / 2

    colortriangulo(2).Alpha = 0

    colortriangulo(2).Blue = ColorAzul(&HFFFFFF)

    colortriangulo(2).Red = ColorRojo(&HFFFFFF)

    colortriangulo(2).Green = ColorVerde(&HFFFFFF)

Vean que el X e Y no es mas que el centro del formulario, y el color que uso es &HFFFFFF que es el blanco.

Luego de configurar nuestros triángulos con sus respectivos vértices llamamos a la función:

GradientFillTriangle Me.hDC, colortriangulo(0), 5, triangulos(0), 4, GRADIENT_FILL_TRIANGLE

Aquí le decimos a Windows lo siguiente: “en el HDC = Me.hDC, dibuja un gradiente cuyo vértices están en el vector “colortriangulo(0)” a su vez le indicamos que son 5, luego le paso el orden de cómo configuro los vértices de cada triangulo en la variable “triángulos”, le indicamos que configuramos 4 triángulos, y al final le indicamos que llene con el degrado cada triangulo.”

¿Lo ven? Puede ser difícil al principio, para mi lo fue, es interesante ver que algo que podrías considerar relativamente simple, requiera de muchas cositas para su funcionamiento, pero queda claro que con esta función se pueden hacer efectos muy interesantes a nivel grafico.

NOTA: aquí utilizamos 2 funciones API, una para el degrado rectangular y otra para el triangulo, en caso de usar C++ o C pueden usar una solo función que se llama GradientFill, ¿Cómo sabe si es rectangular o triangular?. Básicamente por el parámetro número 4, “pMesh” y por el último parámetro donde le pasamos las constantes GRADIENT_FILL_RECT_H, GRADIENT_FILL_RECT_V  o GRADIENT_FILL_TRIANGLE. De todas maneras pueden informarse por la ayuda de  la documentación oficial en Internet.

SEMITRANSPARENCIA

A partir de Windows 98, se incluyo una nueva librería que permite poner imágenes encima de otras con cierta semitransparencia, efecto muy interesante en algunas ocasiones. Para este proyecto solo se necesitan 2 pictureBox, déjenlos con los nombre que VB les pone por defecto. Luego coloquen las imágenes Costa3.jpg y Costa4.jpg en la carpeta donde esta el proyecto, al picture1 le colocan Costa3.jpg, al picture2 Costa4.jpg y por último peguen este código en el formulario:

Option Explicit

Private Type BLENDFUNCTION

  BlendOp As Byte

  BlendFlags As Byte

  SourceConstantAlpha As Byte

  AlphaFormat As Byte

End Type

Private Const AC_SRC_OVER = &H0

Private Declare Function AlphaBlend Lib "msimg32.dll" (ByVal hdcDest As Long, ByVal xOriginDest As Long, ByVal yOriginDest As Long, ByVal WidthDest As Long, ByVal HeightDest As Long, ByVal hdcSrc As Long, ByVal xOriginSrc As Long, ByVal yOriginSrc As Long, ByVal WidthSrc As Long, ByVal HeightSrc As Long, ByVal BLENDFUNCT As Long) As Long

Private Declare Sub RtlMoveMemory Lib "kernel32.dll" (Destination As Any, Source As Any, ByVal Length As Long)

Dim blend As BLENDFUNCTION

Dim blendlong As Long

Dim contador As Integer

Private Sub Form_Load()

    Picture1.ScaleMode = vbPixels

    Picture2.ScaleMode = vbPixels

    Picture1.AutoRedraw = True

    Picture2.AutoRedraw = True

    Picture1.Top = 0

    Picture1.Left = 0

    Picture2.Top = 0

    Picture2.Left = 0

    Picture1.AutoSize = True

    Picture2.AutoSize = True

    Me.WindowState = 2

End Sub

Private Sub Picture1_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)

    contador = contador + 5

    blend.BlendOp = AC_SRC_OVER

    blend.BlendFlags = 0

    blend.SourceConstantAlpha = contador

    blend.AlphaFormat = 0

    RtlMoveMemory blendlong, blend, 4

    AlphaBlend Picture1.hDC, 0, 0, Picture1.ScaleWidth, Picture1.ScaleHeight, Picture2.hDC, 0, 0, Picture2.ScaleWidth, Picture2.ScaleHeight, blendlong

    Picture1.Refresh

End Sub

Para ver el efecto, den clic a la imagen que se muestra en el picture que cubre toda la pantalla, a medida que dan clic se dan cuenta que la otra imagen poco a poco se va superponiendo a la original.

AlphaBlend

Api: Declare Function AlphaBlend Lib "msimg32.dll" (ByVal hdcDest As Long, ByVal xOriginDest As Long, ByVal yOriginDest As Long, ByVal WidthDest As Long, ByVal HeightDest As Long, ByVal hdcSrc As Long, ByVal xOriginSrc As Long, ByVal yOriginSrc As Long, ByVal WidthSrc As Long, ByVal HeightSrc As Long, ByVal BLENDFUNCT As Long) As Long

Esta es la función encargada de tal efecto, ella consta de 10 parámetros similares a las funciones antes vistas:

HdcDest = HDC del dispositivo destino.

XOriginDest = Coordenada X de la esquina superior izquierda del rectángulo destino

YOriginDest = Coordenada Y de la esquina superior izquierda del rectángulo destino

WidthDest = Ancho del rectángulo destino.

HeightDest = Alto del rectángulo destino

HdcSrc = HDC del dispositivo origen

XOrigintSrc = Coordenada X de la esquina superior izquierda del rectángulo origen

YOriginSrc = Coordenada Y de la esquina superior izquierda del rectángulo origen

WidthSrc = Ancho del rectángulo origen

HeightSrc = Alto del rectángulo origen.

Estos parámetros ya deben de ser familiares para ustedes, el parámetro donde se configura el efecto de transparencia es el último, BLENDFUNCT que es una estructura.

Private Type BLENDFUNCTION

  BlendOp As Byte

  BlendFlags As Byte

  SourceConstantAlpha As Byte

  AlphaFormat As Byte

End Type
Esta estructura todavía esta en sus inicios. El parámetro BlendOp en al actualidad solo puede tomar un solo parámetro y es AC_SRC_OVER. BlendFlags se le pone 0 ya que todavía no se le ha programado la utilidad que en un futuro deberá cumplir. El parámetro de la estructura que en realidad es importante es SourceConstantAlpha este puede tomar un valor entre 0 y 255, siendo 0 transparencia total, y 255 opacidad total. Jugando con los valores intermedios tenemos la semitransparencia, es decir, podemos ver las dos imágenes al mismo tiempo.

Por último tenemos el parámetro AlphaFormat que en nuestro ejemplo tiene 0, lo que significa que la transparencia es global. En caso de tomar esta constante:

Private Const AC_SRC_ALPHA = &H1

La transparencia sería Local.

¿Qué es eso de transparencia local y global? En nuestro proyecto usamos transparencia global, lo que quiere decir, es que se aplica la misma transparencia a todos los píxeles de la imagen. ¿Qué quiero decir? Si ponen en SourceConstantAlpha = 125 la función procesara la transparencia en todos los píxeles basándose en el valor de 125.

Con la transparencia local no sucede lo mismo, pero en este caso hay que usar imágenes de 32 Bits, no se pueden usar imágenes de 24 Bits. ¿Por qué? Las imágenes de 32 Bits se componen de la siguiente manera 0x00aavvrr, siendo a = azul, v = verde y r = rojo el RGB que ustedes conocen. Pero ahora hay un 00 a la izquierda del azul, ese 00 es la intensidad ALPHA para ese píxel, es decir, la transparencia de un píxel es independiente del otro píxel. Lo que no sucede con la transparencia global. No muestro un ejemplo de ello, ya que carezco de un software grafico que permita crear una imagen de 32 bit donde pueda tocar este byte encargado del ALPHA o transparencia.

Pero en caso de que ustedes lo hagan, la función se escribe exactamente igual con la diferencia que en la estructura BLENDFUNCTION en el parámetro AlphaFormat colocan la constante que mencione arriba. Cabe destacar que al usar Transparencia Local no importa el valor que introduzcan en SourceConstantAlpha, ya que él toma el valor del byte alpha del píxel de la imagen.

En nuestro proyecto yo lo que hago es incrementar SourceConstantAlpha con una variable “contador” cada vez que se presiona clic en la imagen. Este contador empieza desde 0 hasta 255 que es cuando se ve la segunda imagen completa.

¿Si el último parámetro es una estructura por que en la función aparece como un tipo Long? En realidad no se por que el valor es de tipo long, pero lo curioso no es la estructura, es esta función API:

    RtlMoveMemory vlong, blend, 4    

Esta función API la ayuda de MSND no la tiene, así que no la va explicar tal cual como es, sino que voy hacer una interpretación lógica de la función.

Ya vimos que la función API AlphaBlend no acepta la estructura como parámetro sino un valor de tipo “long” (Estoy hablando del parámetro número 11), lo que yo entiendo que hace esta API es convertir la estructura a una variable de tipo LONG (NOTA: estoy suponiendo que la función API realiza eso, ya que no encontré en la ayuda de MSDN la definición de la misma), ya que si vemos el API en su declaración:

Private Declare Sub RtlMoveMemory Lib "kernel32.dll" (Destination As Any, Source As Any, ByVal Length As Long)


Destination = es “vLogn” variable que declaramos de tipo Long


Source = la estructura blend


Lenght = 4, esto se debe a que la variables de tipo Long son de 4 Bytes.

Luego que convertimos la estructura a una variable de tipo long se la pasamos sin problema a la función API AlphaBlend. Esto de “convertir” depende de cómo se vea, pueda que la variable colocada en “Destination” tenga es un puntero a la estructura.

TRANSPARENCIA

Es usual ver en la televisión esos programas que muestran los trucos para hacer las películas, uno de los cuales es grabar a los actores en un fondo cuyo color sea sólido y luego la computadora suplanta ese color por una imagen ya cargada. Menciono esto, ya que Windows a partir de la versión 98 incluyó una función que hace tal efecto, es decir, dada una imagen con un color fijo de fondo podemos extraer ese color dejando solo la imagen que nos interesa. 

Para este proyecto solo se necesitan 2 pictureBox y un botón, los nombres de cada control serán los que VB les pone por defecto, al “picture1”, le asignan la imagen “transparencia.bmp” que viene con este documento. Y luego pegan este código:

Private Declare Function TransparentBlt Lib "msimg32.dll" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal nSrcWidth As Long, ByVal nSrcHeight As Long, ByVal crTransparent As Long) As Boolean

Private Sub Command1_Click()

    DoEvents

    TransparentBlt Picture2.hdc, 0, 0, Picture1.ScaleWidth, Picture1.ScaleHeight, Picture1.hdc, 0, 0, Picture1.ScaleWidth, Picture1.ScaleHeight, RGB(255, 255, 0)

End Sub

Private Sub Form_Load()

    Command1.Top = 4815

    Command1.Left = 8100

    Command1.Caption = "PRESIONAME"

    Picture1.ScaleMode = 3

    Picture2.ScaleMode = 3

    Picture1.Top = 180

    Picture1.Left = 90

    Picture2.Top = 180

    Picture2.Left = 4770

    Me.Width = 9570

    Me.Height = 5790

End Sub

Como pueden ver es un ejemplo muy sencillo, al correr el programa ven una imagen con un fondo amarillo, luego al presionar clic en el botón, el picture 2 toma la imagen pero sin ese color.

TransparentBlt

Api: Declare Function TransparentBlt Lib "msimg32.dll" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal nSrcWidth As Long, ByVal nSrcHeight As Long, ByVal crTransparent As Long) As Boolean

A estas alturas del manual los 10 primeros parámetros de esta función ya deben de ser familiares para ustedes, los primeros 5 especifican el área del HDC destino que recibirá la imagen, los otros 5 especifican el área del HDC de origen, es decir, quien contiene la imagen original.

El parámetro interesante es el último “crTransparent”, este último parámetro no es mas que el color que tomaremos como transparente, es decir, en nuestro proyecto tenemos en el parámetro RGB(255,255,0) , con esto le indicamos que a la imagen de origen, le quite todos los píxeles que son amarillo. Ustedes en sus imágenes pueden usar cualquier color como transparente, eso quedara a criterio de cada uno.

Capítulo  V.  Ventana (Windows)

En la introducción y en los capítulos ya explicados hemos hablado sobre ventana, pero en este Capítulo  trabajaremos un poco mas profundo con este concepto. ¿Se recuerdan del código hecho en C?, ese código lo vamos a repetir pero en VB, y explicaremos con un poco mas de detalles su funcionamiento.

Ya sabemos que la mayoría de las cosas que vemos son ventanas, pero hablemos ahora de esas ventanas o formularios que nos sirven para mostrar información. Cuando ustedes arrancan una aplicación esta tiene que correr en una ventana principal (main window), esta ventana principal tiene básicamente la siguiente forma:








Si le dice ”área cliente”, al área donde nosotros podemos dibujar, donde ponemos los controles, etc, en pocas palabras, es nuestro hdc, el área no cliente es aquella en donde Windows es el que se encarga de crearla, como pueden ver nosotros en ningún momento de la creación de la misma especificamos colores para la barra de titulo, en ningún momento pusimos botones de cerrar, maximizar o minimizar, ni le indicamos que pusiera un menú de sistema, por cierto, menú del sistema es el típico menú que se usa, para mover, maximizar, minimizar, etc. Toda aplicación de Windows básicamente tiene lo que mencione, el uso de los Scroll normalmente es opcional y dependerá de la aplicación que estemos desarrollando.

La figura mostrada sería una ventana “básica”, no tiene por que ser así obligatoriamente, las ventanas están compuestas o se crean dependiendo de los “estilos” que usemos, cuando creamos una ventana nosotros configuramos ciertos estilos que pueden cambiar la apariencia de nuestra ventana, estos estilos se presentan a continuación:

Public Const WS_OVERLAPPED = &H0& : Con este estilo se crea una ventana con bordes y barra de títulos o caption, se llama caption a la banda coloreada con el titulo de la ventana.

Public Const WS_POPUP = &H80000000 : Con este estilo, se crea una ventana sin bordes, sin caption, nada prácticamente, lo único que se muestra es el área del cliente.

Public Const WS_CHILD = &H40000000 : Sirve para crear ventanas que son hijas.

Public Const WS_MINIMIZE = &H20000000 : Crea una ventana que inicialmente se encuentra minimizada, este estilo solo se usa con WS_OVERLAPPED.

Public Const WS_VISIBLE = &H10000000 : Crea una ventana que es visible.

Public Const WS_DISABLED = &H8000000 : Crea una ventana que esta desactivada.

Public Const WS_MAXIMIZE = &H1000000 : Crea una ventana con tamaño maximizado.

Public Const WS_CAPTION = &HC00000  : Crea una ventana con la barra de titulo, y activa el borde de la misma

Public Const WS_BORDER = &H800000 : Crea una ventana con borde

Public Const WS_DLGFRAME = &H400000 : Crea una ventana con doble borde pero sin titulo

Public Const WS_VSCROLL = &H200000 : Crea una ventana con una barra de desplazamiento vertical.

Public Const WS_HSCROLL = &H100000  : Crea una ventana con una barra de desplazamiento horizontal.

Public Const WS_SYSMENU = &H80000 : Crea una ventana con el menú del sistema.

Public Const WS_THICKFRAME = &H40000 : Crea una ventana con bordes, los cuales pueden servir para redimensionar la ventana.

Public Const WS_GROUP = &H20000 : Estilo usado cuando trabajamos con grupos.

Public Const WS_TABSTOP = &H10000 : Crea una ventana que puede tomar el foco cuando se presiona la tecla TAB

Public Const WS_MINIMIZEBOX = &H20000 : Crea una ventana con el botón de minimizar

Public Const WS_MAXIMIZEBOX = &H10000 : Crea una ventana con el botón de maximizar

Public Const WS_TILED = WS_OVERLAPPED 

Public Const WS_ICONIC = WS_MINIMIZE

Public Const WS_SIZEBOX = WS_THICKFRAME

Public Const WS_OVERLAPPEDWINDOW = (WS_OVERLAPPED Or WS_CAPTION Or WS_SYSMENU Or WS_THICKFRAME Or WS_MINIMIZEBOX Or WS_MAXIMIZEBOX)

Public Const WS_TILEDWINDOW = WS_OVERLAPPEDWINDOW

Public Const WS_POPUPWINDOW = (WS_POPUP Or WS_BORDER Or WS_SYSMENU)

Public Const WS_CHILDWINDOW = (WS_CHILD)

Como verán los últimos estilos son  iguales o combinaciones de otros estilos. Con estos estilos se pueden crear la ventana que ustedes quieran.

INSTANCIA (Instance handle)

Cuando una aplicación arranca, a este se le asigna una “instancia”, que es un número que usa el sistema operativo para saber que funciones o que bloque de información, son de esa aplicación. ¿Qué quiero decir?. Imaginen que tenemos un programa con 4 funciones, al arrancar el programa estas funciones son colocadas en memoria, pero se puede arrancar el mismo programa de nuevo, es decir, se inicia otra instancia, este otro programa tiene las mismas funciones que el primero, pero Windows sabe como direccionar el llamado de las funciones gracias a la instancia creada.

PADRES  E HIJOS (PARENT – CHILD)

Si tenemos un botón en un formulario, este botón es “HIJO” del formulario, o esta contenido en él. Es decir, se podría decir, con son hijos todas aquellas ventanas contenidas en otra, la cual es llamada “Parent” o padre. La ventana principal o “Main Window”, es nuestro padre principal, en él nosotros incluimos otras ventanas que son “hijos”, esto implica que al destruir al padre todas las ventanas hijas que están contenidas en él también son destruidas.

Una pregunta podría ser ¿La ventana principal es hija de alguien?. SI!, del escritorio de Windows o “Desktop Windows”, se puede ver al escritorio de Windows como el padre supremo. ¿Escritorio de Windows, que es eso?. Cuando arrancas Windows, él inicia un programa llamado “explorer.exe”, gracias a ese programa, ustedes ven la barra de inicio (donde esta el botón “inicio”), ven los iconos, y las diferentes ventanas  que Windows le proporcionan, si no me creen aprieten CTRL.-ALT-DEL y finalicen el programa “explorer” verán que se desaparece todo. 

Esto que es sencillo de entender, deben de tenerlo en claro al momento de crear una ventana ya que en su creación habría que determinar si la ventana es hija de alguien o es una ventana principal.

CLASE (CLASS)
Una clase, es como una plantilla, en esta plantilla podemos encontrar valores u objetos que toma una ventana al momento de su creación. Windows proporciona funciones para crear nuestras propias clases, inclusive cuando creamos una ventana usualmente se crea una clase. Esto es un poco mas complicado,  los botones, por ejemplo, Windows los tiene en una clase “button”, es decir, Windows nos da una plantilla con ciertos parámetro ya predefinidos, con esa plantilla nosotros creamos una ventana que tiene la forma de un botón.

La clase se configura a través de una estructura:

Type WNDCLASS

    style As Long

    lpfnwndproc As Long

    cbClsextra As Long

    cbWndExtra2 As Long

    hInstance As Long

    hIcon As Long

    hCursor As Long

    hbrBackground As Long

    lpszMenuName As String

    lpszClassName As String

End Type

Style = aquí especificamos el estilo de la clase, OJO estilo de la clase, no estilo de la ventana, en la clase no se configuran los parámetros de cómo va ser la ventana, eso lo veremos mas adelante.

Entre los valores que puede tomar son los siguiente:

Public Const CS_BYTEALIGNCLIENT = &H1000

Public Const CS_BYTEALIGNWINDOW = &H2000

Public Const CS_CLASSDC = &H40

Public Const CS_DBLCLKS = &H8

Public Const CS_GLOBALCLASS = &H4000

Public Const CS_HREDRAW = &H2

Public Const CS_NOCLOSE = &H200

Public Const CS_OWNDC = &H20

Public Const CS_PARENTDC = &H80

Public Const CS_SAVEBITS = &H800

Public Const CS_VREDRAW = &H1

No les voy a mentir, los único que se usar son CS_VREDRAW y CS_HREDRAW, estos estilos, lo que hacen es que cuando la ventana es movida o alterada en su tamaño ya sea horizontal o vertical, la ventana es repintada, es decir, se envía un mensaje WM_PAINT, indicando que la ventana tiene que ser repintada.

Para las otras constantes que en realidad no le he visto uso, pueden documentarse en la ayuda de MSDN.

Lpfnwndproc = Aquí colocamos la dirección de nuestro “oído” o la función que gestiona los mensajes de la ventana, en la documentación oficial el gestor de mensajes lo pueden ver como “Windows Procedure”.

CbClsextra y cbWndExtra2 = tampoco le veo uso a estas dos variables de la estructura, según la documentación lo que hacen es asignar memoria extra. Pero en nuestros ejemplos vamos a poner 0 en ambas.

HInstance = instancia de la aplicación que contiene el gestor de mensajes.

HIcon = Icono que se muestra en la esquina superior izquierda de la ventana.

HCursor = Cursor que se muestra en el momento que el mouse entra en la ventana.

HbrBackground = fondo de la ventana cuando ella es creada.

LpszMenuName = handle del menú que utiliza la ventana.

LpszClassName = nombre de la clase.

Después de configurar como va ser nuestra clase procedemos a registrarla.

RegisterClass

Api: Declare Function RegisterClass Lib "user32" Alias "RegisterClassA" (Class As WNDCLASS) As Long

Esta función se encarga de registrar una clase, pueden ver que como único parámetro se le pasa una variable del tipo WNDCLASS.

Si la función retorna 0 significa que la creación de la clase fallo. En caso de ser exitosa la creación retornara un número distinto de 0.

RegisterClassEx

Api: Declare Function RegisterClassEx Lib "user32" Alias "RegisterClassExA" (pcWndClassEx As WNDCLASSEX) As Integer

Hace exactamente lo mismo que la función API anterior con la diferencia que recibe una estructura del tipo WNDCLASSEX.

Esta estructura es casi igual a WNDCLASS:

Type WNDCLASSEX


cbSize As Long


style As Long


lpfnWndProc As Long


cbClsExtra As Long


cbWndExtra As Long


hInstance As Long


hIcon As Long


hCursor As Long


hbrBackground As Long


lpszMenuName As String


lpszClassName As String


hIconSm As Long

End Type

Tiene solo dos variables nuevas, cbSize que es el tamaño de la estructura, en VB pueden usar la función Len() para ello y en C++ se usa “sizeof”. Y la variable “hIconSm” que no es mas que el icono mostrado en la barra inferior cuando la ventana se minimiza. De resto es exactamente lo mismo.

UnregisterClass

Api: Declare Function UnregisterClass Lib "user32" Alias "UnregisterClassA" (ByVal lpClassName As String, ByVal hInstance As Long) As Long

Esta función lo que hace es eliminar la clase registrada. Como primer parámetro se le pasa el nombre de la clase, y luego la instancia de la aplicación.

Si esto todavía no lo captan esperen ver el ejemplo que se presenta a continuación.

VENTANA NUEVA

Para este proyecto solo es necesario un modulo, es decir, no tiene formulario. En el modulo pongan este código, y configuren en “PROYECTO- Propiedades del proyecto...” la opción de arrancar por el “Sub main”.

Código:

Declare Function RegisterClass Lib "user32" Alias "RegisterClassA" (Class As WNDCLASS) As Long

Declare Function UnregisterClass Lib "user32" Alias "UnregisterClassA" (ByVal lpClassName As String, ByVal hInstance As Long) As Long

Declare Function CreateWindowEx Lib "user32" Alias "CreateWindowExA" (ByVal dwExStyle As Long, ByVal lpClassName As String, ByVal lpWindowName As String, ByVal dwStyle As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hWndParent As Long, ByVal hMenu As Long, ByVal hInstance As Long, lpParam As Any) As Long

Declare Function DefWindowProc Lib "user32" Alias "DefWindowProcA" (ByVal hWnd As Long, ByVal wMsg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Declare Sub PostQuitMessage Lib "user32" (ByVal nExitCode As Long)

Declare Function GetMessage Lib "user32" Alias "GetMessageA" (lpMsg As Msg, ByVal hWnd As Long, ByVal wMsgFilterMin As Long, ByVal wMsgFilterMax As Long) As Long

Declare Function TranslateMessage Lib "user32" (lpMsg As Msg) As Long

Declare Function DispatchMessage Lib "user32" Alias "DispatchMessageA" (lpMsg As Msg) As Long

Declare Function ShowWindow Lib "user32" (ByVal hWnd As Long, ByVal nCmdShow As Long) As Long

Declare Function LoadCursor Lib "user32" Alias "LoadCursorA" (ByVal hInstance As Long, ByVal lpCursorName As Any) As Long

Declare Function LoadIcon Lib "user32" Alias "LoadIconA" (ByVal hInstance As Long, ByVal lpIconName As Long) As Long

Declare Function DestroyWindow Lib "user32" (ByVal hWnd As Long) As Long

Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Public Type WNDCLASSEX

    cbSize As Long

    style As Long

    lpfnWndProc As Long

    cbClsExtra As Long

    cbWndExtra As Long

    hInstance As Long

    hIcon As Long

    hCursor As Long

    hbrBackground As Long

    lpszMenuName As String

    lpszClassName As String

    hIconSm As Long

End Type

Type WNDCLASS

    style As Long

    lpfnWndProc As Long

    cbClsExtra As Long

    cbWndExtra2 As Long

    hInstance As Long

    hIcon As Long

    hCursor As Long

    hbrBackground As Long

    lpszMenuName As String

    lpszClassName As String

End Type

Type POINTAPI

    x As Long

    y As Long

End Type

Type Msg

    hWnd As Long

    message As Long

    wParam As Long

    lParam As Long

    time As Long

    pt As POINTAPI

End Type

Public Const WS_OVERLAPPED = &H0&

Public Const WS_POPUP = &H80000000

Public Const WS_CHILD = &H40000000

Public Const WS_MINIMIZE = &H20000000

Public Const WS_VISIBLE = &H10000000

Public Const WS_DISABLED = &H8000000

Public Const WS_CLIPSIBLINGS = &H4000000

Public Const WS_CLIPCHILDREN = &H2000000

Public Const WS_MAXIMIZE = &H1000000

Public Const WS_CAPTION = &HC00000                  '  WS_BORDER Or WS_DLGFRAME

Public Const WS_BORDER = &H800000

Public Const WS_DLGFRAME = &H400000

Public Const WS_VSCROLL = &H200000

Public Const WS_HSCROLL = &H100000

Public Const WS_SYSMENU = &H80000

Public Const WS_THICKFRAME = &H40000

Public Const WS_GROUP = &H20000

Public Const WS_TABSTOP = &H10000

Public Const WS_MINIMIZEBOX = &H20000

Public Const WS_MAXIMIZEBOX = &H10000

Public Const WS_TILED = WS_OVERLAPPED

Public Const WS_ICONIC = WS_MINIMIZE

Public Const WS_SIZEBOX = WS_THICKFRAME

Public Const WS_OVERLAPPEDWINDOW = (WS_OVERLAPPED Or WS_CAPTION Or WS_SYSMENU Or WS_THICKFRAME Or WS_MINIMIZEBOX Or WS_MAXIMIZEBOX)

Public Const WS_TILEDWINDOW = WS_OVERLAPPEDWINDOW

Public Const WS_POPUPWINDOW = (WS_POPUP Or WS_BORDER Or WS_SYSMENU)

Public Const WS_CHILDWINDOW = (WS_CHILD)

Public Const WS_EX_CONTEXTHELP = &H400&

Public Const BS_PUSHBUTTON = &H0&

Public Const CS_VREDRAW = &H1

Public Const CS_HREDRAW = &H2

Public Const SW_NORMAL = 1

Public Const WM_CREATE = &H1

Public Const WM_DESTROY = &H2

Public Const WM_HELP = &H53

Public Const WM_LBUTTONUP = &H202

Public Const WM_COMMAND = &H111

Public Const IDI_APPLICATION = 32512&

Public Const IDC_ARROW = 32512&

Dim hwnd2 As Long, hwnd3 As Long, hwndHija As Long

Dim hWnd As Long

Public Sub Main()

    Dim mensaje As Msg

    If Not RegistrarClase(AddressOf GestorMensajes) Then

        MsgBox "Falla en la creacion de la clase"

        UnregisterClass "mipropiaclase", App.hInstance

        Exit Sub

    End If

    If Not CrearAplicacion Then

        MsgBox "Falla en la creacion de la aplicacion"

        UnregisterClass "mipropiaclase", App.hInstance

        Exit Sub

    End If

    Do While (GetMessage(mensaje, 0, 0, 0))

        TranslateMessage mensaje

        DispatchMessage mensaje

    Loop

    UnregisterClass "mipropiaclase", App.hInstance

End Sub

Private Function RegistrarClase(FuncionMensajes As Long) As Boolean

    Dim clase As WNDCLASS

    clase.style = CS_HREDRAW Or CS_VREDRAW

    clase.lpfnWndProc = FuncionMensajes

    clase.cbClsExtra = 0

    clase.cbWndExtra2 = 0

    clase.hInstance = App.hInstance

    clase.hIcon = LoadIcon(0, IDI_APPLICATION)

    clase.hCursor = LoadCursor(0, IDC_ARROW)

    clase.hbrBackground = CreateSolidBrush(RGB(255, 255, 255))

    clase.lpszMenuName = 0

    clase.lpszClassName = "mipropiaclase"

    RegistrarClase = (RegisterClass(clase) <> 0)

End Function

Private Function CrearAplicacion() As Boolean

    'Tipos de Ventanas Principales

    hWnd = CreateWindowEx(0, "mipropiaclase", "Ventana Principal", WS_OVERLAPPEDWINDOW Or WS_HSCROLL Or WS_VSCROLL, 0, 0, 500, 400, 0, 0, App.hInstance, ByVal 0&)

    'hwnd = CreateWindowEx(0, "mipropiaclase", "Ventana Principal", WS_CAPTION, 0, 0, 400, 300, 0, 0, App.hInstance, ByVal 0&)

    'hwnd = CreateWindowEx(0, "mipropiaclase", "Ventana Principal", WS_POPUP, 0, 0, 400, 300, 0, 0, App.hInstance, ByVal 0&)

    'hwnd = CreateWindowEx(0, "mipropiaclase", "Ventana Principal", WS_POPUPWINDOW, 0, 0, 400, 300, 0, 0, App.hInstance, ByVal 0&)

    'hwnd = CreateWindowEx(0, "mipropiaclase", "Ventana Principal", WS_SIZEBOX, 0, 0, 400, 300, 0, 0, App.hInstance, ByVal 0&)

    'hwnd = CreateWindowEx(0, "mipropiaclase", "Ventana Principal", WS_SIZEBOX Or WS_SYSMENU, 0, 0, 400, 300, 0, 0, App.hInstance, ByVal 0&)

    'hWnd = CreateWindowEx(WS_EX_CONTEXTHELP, "mipropiaclase", "Ventana Principal", WS_SIZEBOX Or WS_SYSMENU, 0, 0, 400, 300, 0, 0, App.hInstance, ByVal 0&)

    'Ventanas Hijas

    hwnd2 = CreateWindowEx(0, "button", "Presioname", BS_PUSHBUTTON Or WS_CHILD, 0, 0, 100, 50, hWnd, 0, App.hInstance, ByVal 0&)

    hwnd3 = CreateWindowEx(0, "button", "Salir", BS_PUSHBUTTON Or WS_CHILD, 0, 100, 100, 50, hWnd, 0, App.hInstance, ByVal 0&)

    If hWnd = 0 Then

        CrearAplicacion = False

        Exit Function

    End If

    ShowWindow hWnd, SW_NORMAL

    ShowWindow hwnd2, SW_NORMAL

    ShowWindow hwnd3, SW_NORMAL

    CrearAplicacion = True

End Function

Private Function GestorMensajes(ByVal hWnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    Select Case mensaje

        Case WM_COMMAND

            Select Case lParam

            Case hwnd2

                hwndHija = CreateWindowEx(0, "mipropiaclase", "Ventana Hija", WS_CHILD Or WS_BORDER Or WS_CAPTION, 200, 100, 300, 300, hWnd, 0, App.hInstance, ByVal 0&)

                ShowWindow hwndHija, SW_NORMAL

            Case hwnd3

                DestroyWindow hWnd

            End Select

        Case WM_DESTROY

            PostQuitMessage (0)

    End Select

    GestorMensajes = DefWindowProc(hWnd, mensaje, wParam, lParam)

End Function
Cuando arranquen verán una ventana, con fondo blanco y dos botones. Aunque este ejemplo es igual al escrito en C al inicio del manual, voy a volver a explicar paso a paso el código, para dejar en claro ciertas funciones.

Empecemos en el Sub Main, con este bloque de código:

    If Not RegistrarClase(AddressOf GestorMensajes) Then

        MsgBox "Falla en la creacion de la clase"

        UnregisterClass "mipropiaclase", App.hInstance

        Exit Sub

    End If

Lo primero que hacemos es llamar a la función RegistrarClase, ya que una ventana NO puede ser creada sin referenciarla a una clase.

Private Function RegistrarClase(FuncionMensajes As Long) As Boolean

    Dim clase As WNDCLASS

    clase.style = CS_HREDRAW Or CS_VREDRAW

    clase.lpfnWndProc = FuncionMensajes

    clase.cbClsExtra = 0

    clase.cbWndExtra2 = 0

    clase.hInstance = App.hInstance

    clase.hIcon = LoadIcon(0, IDI_APPLICATION)

    clase.hCursor = LoadCursor(0, IDC_ARROW)

    clase.hbrBackground = CreateSolidBrush(RGB(255, 255, 255))

    clase.lpszMenuName = 0

    clase.lpszClassName = "mipropiaclase"

    RegistrarClase = (RegisterClass(clase) <> 0)

End Function

Lo primero que hacemos es crear una variable “clase” que sea del tipo WNDCLASS. Luego llenamos los valores, en el parámetro “lpfnWndProc” le asignamos el lugar de memoria en donde se encuentra nuestro “oido” o gestor de mensajes. Los demas parámetro, no deberían de tener problema de entenderlos, ya que el significado de estos parámetro fueron estudiados arriba.

Pero de seguro, algunas personas pueden preguntarse ¿Referenciamos el gestor de mensajes sin crear la ventana?. Recuerden que lo que estamos haciendo es una plantilla, veámoslos con un ejemplo, imaginen un control “X”, ese control “X” posee un gestor de mensajes, el cual se configura en su clase, cada vez que creamos una ventana referenciada a esa clase, el control que se crea , se podría decir, copia el gestor de mensajes que posee la clase, si tenemos 100 controles pues ellos toman 1 gestor de mensaje para cada uno, es decir sería 100 gestores de mensaje.

Imaginen ahora que no existe esto de crear una plantilla o clase, y necesitan crear 100 controles “X”, tendrían que programar “individualmente” 100 gestores de mensajes, esto haría que un control tenga un comportamiento distinto al otro, siendo el mismo control “X”, ahora ven un poco mas claro la utilidad de crear una clase o plantilla antes de crear la ventana.

Esta clase que nosotros creamos la llamamos “mipropiaclase”. Ya registrada la clase podemos entonces crear la ventana:

    If Not CrearAplicacion Then

        MsgBox "Falla en la creacion de la aplicacion"

        UnregisterClass "mipropiaclase", App.hInstance

        Exit Sub

    End If

Este IF llama a la función CrearAplicacion la cual se muestra a continuación:

Private Function CrearAplicacion() As Boolean

    'Tipos de Ventanas Principales

    hWnd = CreateWindowEx(0, "mipropiaclase", "Ventana Principal", WS_OVERLAPPEDWINDOW Or WS_HSCROLL Or WS_VSCROLL, 0, 0, 500, 400, 0, 0, App.hInstance, ByVal 0&)


.


.


.

    'Ventanas Hijas

    hwnd2 = CreateWindowEx(0, "button", "Presioname", BS_PUSHBUTTON Or WS_CHILD, 0, 0, 100, 50, hWnd, 0, App.hInstance, ByVal 0&)

    hwnd3 = CreateWindowEx(0, "button", "Salir", BS_PUSHBUTTON Or  WS_CHILD, 0, 100, 100, 50, hWnd, 0, App.hInstance, ByVal 0&)

    If hWnd = 0 Then

        CrearAplicacion = False

        Exit Function

    End If

    ShowWindow hWnd, SW_NORMAL

    ShowWindow hwnd2, SW_NORMAL

    ShowWindow hwnd3, SW_NORMAL

    CrearAplicacion = True

End Function

En esta función es donde procedemos a crear nuestra ventana, para ello expliquemos la siguiente función:

CreateWindowEx

Api: Declare Function CreateWindowEx Lib "user32" Alias "CreateWindowExA" (ByVal dwExStyle As Long, ByVal lpClassName As String, ByVal lpWindowName As String, ByVal dwStyle As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hWndParent As Long, ByVal hMenu As Long, ByVal hInstance As Long, lpParam As Any) As Long

Esta función sirve para crear “fisicamente” la ventana. Los parámetros son:

DwExStyle: es donde se configuran estilos “extras”, estos valores se pueden combinar usando el operador OR, los valores pueden ser:

Public Const WS_EX_ACCEPTFILES = &H10& : permite que la ventana creada acepte el arrastre o “drap-drop” de archivos.

Public Const WS_EX_APPWINDOW = &H40000& : Fuerza la ventana a ubicarse en el TOP-LEVEL,  cuando ella es visible, esto ocurre al presionar la clic en el “TaskBar”, TaskBar en la barra de botones que se muestra en la barra inferior, donde esta el botón de “inicio”.

Public Const WS_EX_CLIENTEDGE = &H200& 

Public Const WS_EX_CONTEXTHELP = &H400& : Crea una ventana con un botón de “?” en el área no cliente. Este botón envía un mensaje WM_HELP a la ventana hija que se encuentre en la ventana en el momento de presionar el clic del mouse. Este estilo no puede ser usado con los estilos de ventanas WM_MAXIMIZEBOX y WM_MINIMIZEBOX.

Public Const WS_EX_CONTROLPARENT = &H10000& : Permite a la ventana navegar entre las ventanas hijas por medio de la tecla TAB.

Public Const WS_EX_DLGMODALFRAME = &H1& : Crea una ventana con doble borde

Public Const WS_EX_LEFT = &H& : Crea una ventana que posee una alineación hacia la izquierda, este estilo esta por defecto.

Public Const WS_EX_LEFTSCROLLBAR = &H4000&

Public Const WS_EX_LTRREADING = &H& : Ventana cuyo texto desplegado se representa de izquierda a derecha, este estilo también esta por defecto.

Public Const WS_EX_MDICHILD = &H40& : crea una ventana hija MDI.

Public Const WS_EX_NOPARENTNOTIFY = &H4& : Especifica que la ventana hija no envía el mensaje WM_PARENTNOTIFY cuando esa ventana es creada o destruida.

Public Const WS_EX_OVERLAPPEDWINDOW = (WS_EX_WINDOWEDGE Or WS_EX_CLIENTEDGE)

Public Const WS_EX_PALETTEWINDOW = (WS_EX_WINDOWEDGE Or WS_EX_TOOLWINDOW Or WS_EX_TOPMOST)

Public Const WS_EX_RIGHT = &H1000& : Ventana cuya propiedad de alineación es “derecha”

Public Const WS_EX_RIGHTSCROLLBAR = &H& : Si existe una barra de desplazamiento vertical esta se encuentra a la derecha de la ventana. Estilo por defecto.

Public Const WS_EX_RTLREADING = &H2000& : El texto desplegado en la ventana se representa de derecha – izquierda, utilizado para los casos en que se representan texto con lenguajes arabes.

Public Const WS_EX_STATICEDGE = &H20000& : crea una ventana con un borde tridimensional, pensado para ser usado en elementos que no aceptan entrada de usuario.

Public Const WS_EX_TOOLWINDOW = &H80& 

Public Const WS_EX_TOPMOST = &H8& : Fuerza a la ventana a ser ubicada en el TOP-LEVEL, inclusive si esta desactivada se mantiene por encima de todas las demas.

Public Const WS_EX_TRANSPARENT = &H20& 

Public Const WS_EX_WINDOWEDGE = &H100& : Crea una ventana con un borde un poco levantado

Algunas constantes no fueron explicadas, una por que son combinaciones de otras ya existentes, y las otras les dejo para que vean se explicación en la documentación oficial, ya que no les veo uso, desde mi punto de vista.

LpClassName = en este parámetro pasamos el nombre de la clase que fue registrada previamente.

LpWindowName = le pasamos el nombre de la ventana, en un formulario equivale al nombre que es mostrado en el CAPTION.

DwStyle: estilo de la ventana, estos estilos ya fueron explicados arriba, y se pueden combinar con el operador OR.

x, y, nWidth, nHeight : estos 4 parámetros indican la posición de la ventana en la pantalla y su respectivo tamaño, los valores vienen especificados en píxeles.

HWndParent = en caso de ser una ventana hija, aquí es donde se coloca el hwnd de la ventana padre.

HMenu = en caso de poseer menú, es aquí donde se coloca el handle del objeto menú

HInstance = le pasamos la instancia de la aplicación.

LpParam = utilizada para pasarle una variable del tipo CREATESTRUCT, pero en realidad todavía yo no le he visto utilidad a esta variable.

Si la función tiene éxito al crear la ventana retorna un valor distinto de 0.

¿Como la usamos?. Veamos nuestro ejemplo:

hWnd = CreateWindowEx(0, "mipropiaclase", "Ventana Principal", WS_OVERLAPPEDWINDOW Or WS_HSCROLL Or WS_VSCROLL, 0, 0, 500, 400, 0, 0, App.hInstance, ByVal 0&)

Con lo explicado, ya deberían de entender lo que hace la función, básicamente estamos creando una ventana que no es hija, ya que en HwndParent estamos pasando 0, pero OJO, no es hija de una ventana creada por nosotros, pero es hija del Desktop Windows o Escritorio de Windows. Noten que como segundo parámetro le pasamos el nombre de la clase registrada por nosotros con anterioridad.

En el código puse ejemplos de varias ventanas, véanlas una por una, quitando los comentarios correspondiente.

Vamos a ver estas dos líneas:
    hwnd2 = CreateWindowEx(0, "button", "Presioname", BS_PUSHBUTTON Or WS_CHILD, 0, 0, 100, 50, hWnd, 0, App.hInstance, ByVal 0&)

    hwnd3 = CreateWindowEx(0, "button", "Salir", BS_PUSHBUTTON Or  WS_CHILD, 0, 100, 100, 50, hWnd, 0, App.hInstance, ByVal 0&)

Con estas dos líneas creamos 2 botones, de la clase “button”, esta clase no fue creada por nosotros; Windows tiene unas clases “system class” o clases de sistemas, las cuales sirven para la creación de controles básicos como un botón, combobox, listbox, edit, o label (static). Ya estas clases están registradas por Windows cuando este arranca.

Pero vean lo siguiente, estas ventanas son hijas de la ventana principal, como son hijas hay que especificar WS_CHILD en los estilos y un “hwnd” en el parámetro HwndParent con esto le estamos diciendo que son hijas del “hwnd”, en pocas palabras son hijas de la ventana que creamos al principio. Hay algo curioso que es la constante BS_PUSHBUTTON, ¿Es esto un estilo de ventana?. No exactamente, este estilo es especifico y solo sirve para los botones, no funciona para ningún otro control. Cada control de las clases del sistema “system class”, posee estilos propios, el documentarse sobre esto, si se los dejo a ustedes, pero no se preocupen no son nada del otro mundo, si han entendido mas o menos todo esto de seguro entenderán esos estilos sin problemas.

Lo que sucede es que no quiero extender el Capítulo  en cosas muy pocos usadas. Pero ustedes pueden decir ¿Para que sirve todo lo visto hasta el momento?. Al final del Capítulo  les contare una anécdota.

Luego de haber creado las ventanas hay que mostrarlas y para ello es la función que se muestra a continuación:

ShowWindow

Api: Declare Function ShowWindow Lib "user32" (ByVal hWnd As Long, ByVal nCmdShow As Long) As Long

Esta función realiza una acción dependiendo del valor del parámetro nCmdShow, pueden ver que como primer parámetro se le pasa el “hwnd” o ventana a la cual queremos aplicar dicha acción.

Valores para nCmdShow:

Public Const SW_HIDE = 0 : Esconde la ventana 

Public Const SW_MAXIMIZE = 3 : Maximiza la ventana especificada.

Public Const SW_MINIMIZE = 6 : Minimiza la ventana especificada.

Public Const SW_RESTORE = 9 : Restaura la ventana, si esta se encuentra maximizada o minimizada, es parecido a lo que hace el botón que esta entre el boton de minimizar y maximizar.

Public Const SW_SHOW = 5 : Activa y muestra la ventana, usando el tamaño y posición actual.

Public Const SW_SHOWMAXIMIZED = 3 : Activa la ventana y la maximiza.

Public Const SW_SHOWMINIMIZED = 2 : Activa la ventana y la minimiza.

Public Const SW_SHOWMINNOACTIVE = 7 : Muestra la ventana minimizada, y no la activa, dejando activa la que se encuentra en esa momento activada.

Public Const SW_SHOWNA = 8 : Muestra la ventana pero sin activarla, es decir, sin darle el foco.

Public Const SW_SHOWNOACTIVATE = 4 : Igual que SW_SHOW pero no la activa.

Public Const SW_SHOWNORMAL = 1 : Activa y muestra la ventana.

Pueden ver que es sencillo lo que hace esta función.

Luego de mostrar las ventanas en caso de que hayan tenido éxito su creación, nos vamos a este bloque de código:

    Do While (GetMessage(mensaje, 0, 0, 0))

        TranslateMessage mensaje

        DispatchMessage mensaje

    Loop

Este bucle “finito” es el que habilita nuestro oido para escuchar.

GetMessage

Api: Declare Function GetMessage Lib "user32" Alias "GetMessageA" (lpMsg As Msg, ByVal hWnd As Long, ByVal wMsgFilterMin As Long, ByVal wMsgFilterMax As Long) As Long

Función que toma el mensaje que se encuentra en la cola. Como primer parámetro le pasamos una variable que es de tipo MSG, la cual es una estructura con las siguientes características:

Type Msg

    hWnd As Long ‘ Ventana a la cual pertenece el mensaje
    message As Long ‘ Mensaje que se envía
    wParam As Long ‘ Información del mensaje
    lParam As Long ‘ Información del mensaje
    time As Long ‘ Tiempo que tiene el mensaje en la cola.
    pt As POINTAPI ‘ Coordenada del puntero del mouse al momento de poner el mensaje en la cola
End Type

Como segundo parámetro tenemos “hWnd”, que no es mas que la ventana de los cuales queremos recibir sus mensajes, es decir, si colocamos nuestro “hwnd” obtenido en la creación de la ventana, pues el GetMessage recibirá los mensajes destinados solamente a ese HWND. Ustedes dirán ¿Pero pusiste 0, que significa eso?. Que tomo los mensajes correspondiente al hilo de la aplicación.

Todo aplicación cuando arranca, crea lo que se llama un hilo o thread, en ese hilo nosotros ejecutamos todo nuestro código, al poner 0, le digo al programa que tome los mensajes de todas las ventanas  que pertenecen a ese hilo. Si nada mas pongo el “hwnd” que me retorna CreateWindowEx, entonces recibiré los mensajes de la ventana principal, pero los mensajes de cada boton no los recibo, al poner 0, como los botones fueron creados en el hilo de la aplicación, esto significa que la función tomara los mensajes correspondiente a los botones. La parte de hilo o thread será explicado mas adelante en un Capítulo .

WmsgFilterMin, wMsgFilterMax = Para explicar estos parámetros tienen que tener en claro que los mensajes no son mas que números, con estos 2 parámetros podemos filtrar los mensajes que se encuentren en el intervalo que forman estos dos parámetros. Es muy raro utilizar dichos parámetros por lo que usualmente se les coloca 0 para que reciban todos los mensajes sin excepción.

TranslateMessage

Api: Declare Function TranslateMessage Lib "user32" (lpMsg As Msg) As Long

Esta función se encarga de convertir los mensajes que provienen de teclas virtuales en mensajes del tipo CHAR o WM_CHAR., siempre que la tecla tenga su equivalente valor ASCII. Como único parámetro tome un valor del tipo MSG.

DispatchMessage

Api: Declare Function DispatchMessage Lib "user32" Alias "DispatchMessageA" (lpMsg As Msg) As Long

Esta función se encargar de enviar el mensaje al gestor de mensajes configurado para ello. Como único parámetro toma un valor del tipo MSG.

Funciones Adicionales:

LoadCursor

Api: Declare Function LoadCursor Lib "user32" Alias "LoadCursorA" (ByVal hInstance As Long, ByVal lpCursorName As Any) As Long

Función utilizada para cargar un cursor del mouse, como primer parámetro le pasamos la instancia de nuestra aplicación. Y como segundo parámetro le pasamos o el nombre del recurso o una constante que identifica a un cursor que posee  el sistema operativo. Como VB no trabaja con la filosofía de archivos de recursos, nada mas pondré las constante de los cursores predeterminados de Windows. Pero para que la función acepte dicha constante es necesario ponerle 0 al parámetro de hInstance.

Valores para “lpCursorName”:

Public Const IDC_ARROW = 32512& : Muestra la flecha tradicional.

Public Const IDC_APPSTARTING = 32650& : Muestra la flecha tradicional con un pequeño reloj de arena.

Public Const IDC_CROSS = 32515& : Cursor “crusshair”

Public Const IDC_HAND = (32649) : Muestra una manito, este sirve para NT 5.0 en adelante

Public Const IDC_IBEAM = 32513& : Cursor “I-beam”

Public Const IDC_NO = 32648& : Un circulo con una diagonal.

Public Const IDC_SIZEALL = 32646& : Muestra una cruz cuyas puntas son flechas.

Public Const IDC_SIZENESW = 32643& : Muestra una línea diagonal con flechas en sus puntas, en dirección Noreste – suroeste

Public Const IDC_SIZENS = 32645& : Muestra una línea, con flechas en sus puntas, en dirección Norte – sur

Public Const IDC_SIZENWSE = 32642& : Muestra una línea, con flechas en sus puntas, en dirección Noroeste – sureste

Public Const IDC_SIZEWE = 32644&  : Muestra una línea, con flechas en sus puntas, en dirección Oeste – Este

Public Const IDC_UPARROW = 32516& : Flecha vertical.

Public Const IDC_WAIT = 32514& : Reloj de arena.

LoadIcon

Api: Declare Function LoadIcon Lib "user32" Alias "LoadIconA" (ByVal hInstance As Long, ByVal lpIconName As Long) As Long

Al igual que LoadCursor toma como primer parámetro la instancia de la aplicación y el segundo parámetro toma el nombre de un recurso, o una constante que representa a un icono predeterminado del sistema, al igual que LoadCursor en caso de tomar los iconos predeterminados de Windows es necesario poner 0 en el valor de hInstance; esas constantes pueden ser:

Public Const IDI_APPLICATION = 32512& : Icono de la aplicación por defecto

Public Const IDI_ASTERISK = 32516& : Icono ASTERISK

Public Const IDI_ERROR = IDI_HAND 

Public Const IDI_EXCLAMATION = 32515& : Icono de exclamación.

Public Const IDI_HAND = 32513& : Icono de una mano sombreada.

Public Const IDI_INFORMATION = IDI_ASTERISK 

Public Const IDI_QUESTION = 32514& : Icono de un signo de interrogación

Public Const IDI_WARNING = IDI_EXCLAMATION

Public Const IDI_WINLOGO = 32517 : Icono del logo de windows.

DestroyWindow

Api: Declare Function DestroyWindow Lib "user32" (ByVal hWnd As Long) As Long

Función encargada de destruir una aplicación, enviando a la cola de mensajes, el mensaje WM_DESTROY. Recuerden, que si la ventana es la principal, todas las ventanas hijas a él se destruyen.

DefWindowProc

Api: Declare Function DefWindowProc Lib "user32" Alias "DefWindowProcA" (ByVal hWnd As Long, ByVal wMsg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Función encargada de utilizar el gestor de mensajes por defecto de una ventana. ¿Qué quiero decir?. Vamos al ejemplo:

    GestorMensajes = DefWindowProc(hWnd, mensaje, wParam, lParam)

Aquí lo que hacemos, es que el mensaje, sea procesado por el gestor de mensajes por defecto, nuestra aplicación no gestiona todos los mensajes que existen, y siempre es bueno procesarlos todo, pero como algunos no son útiles para nosotros, dejamos que el gestor de mensajes por defecto se encargue de los mensajes que no nos interesa.

PostQuitMessage

Api: Declare Sub PostQuitMessage Lib "user32" (ByVal nExitCode As Long)

Con esta función salimos del hilo de la aplicación, es decir, salimos del Do While (GetMessage(mensaje, 0, 0, 0)), concluyendo así la aplicación. Como parámetro le pasamos 0, y así indicamos la finalización del programa, es como decirle a la aplicación que ya no reciba mas mensajes.

ENUMERANDO VENTANAS

A continuación vamos a ver un proyecto muy interesante, que muestra la manipulación de las ventanas de Windows, lo que vamos hacer en el proyecto es tomar todas las ventanas abiertas, y en algunas vamos hacer ciertas cositas.

Para este proyecto se necesita 1 formulario llamado “frmEnumWin” y 1 modulo. En el formulario colocan lo siguientes controles:

5 Botones.

6 Etiquetas o Label.

2 ListBox

Nota: no aprieten los botones mostrados hasta que yo los indique lo que vamos hacer.

Código del formulario:

Option Explicit

Private Const WM_DESTROY = &H2

Private Const WM_CLOSE = &H10

Private Const BM_CLICK = &HF5

Const CWP_ALL = &H0

Const HWND_TOP = 0

Private Sub Command1_Click()

    EnumChildWindows GetDesktopWindow, AddressOf EnumChildProc, ByVal 0&

End Sub

Private Sub Command2_Click()

    Dim hWnd As String

    hWnd = Mid(List1.Text, 1, InStr(1, List1.Text, " ") - 1)

    SendMessage CLng(hWnd), WM_CLOSE, 0, 0

End Sub

Private Sub Command3_Click()

    Dim hWnd As String

    hWnd = Mid(List2.Text, 1, InStr(1, List2.Text, " ") - 1)

    SetWindowText CLng(hWnd), "6555,"

End Sub

Private Sub Command4_Click()

    Dim hWnd As String

    hWnd = Mid(List2.Text, 1, InStr(1, List2.Text, " ") - 1)

    SendMessage CLng(hWnd), BM_CLICK, 0, 0

End Sub

Private Sub Command5_Click()

    List1.Clear

    List2.Clear

    EnumWindows AddressOf EnumWindowsProc, ByVal 0&

End Sub

Private Sub Form_Load()

    Command1.Left = 5805

    Command1.Top = 270

    Command1.Width = 1815

    Command1.Height = 870

    Command1.Caption = "Enumerar los Child del Destkop de Windows"

    Command2.Left = 180

    Command2.Top = 4050

    Command2.Width = 2715

    Command2.Height = 375

    Command2.Caption = "Destruir Aplicacion"

    Command3.Left = 3060

    Command3.Top = 4050

    Command3.Width = 2715

    Command3.Height = 375

    Command3.Caption = "Enviar Texto"

    Command4.Left = 3060

    Command4.Top = 4545

    Command4.Width = 2715

    Command4.Height = 375

    Command4.Caption = "Enviar &Click"

    Command5.Left = 2610

    Command5.Top = 0

    Command5.Width = 240

    Command5.Height = 240

    Command5.Caption = "R"

    Me.Left = 0

    Me.Top = 0

    Me.Width = 7845

    Me.Height = 5610

    Label1.Left = 180

    Label1.Top = 45

    Label1.Width = 1185

    Label1.Height = 240

    Label1.Caption = "EnumWindow"

    Label2.Left = 3060

    Label2.Top = 45

    Label2.Width = 1185

    Label2.Height = 240

    Label2.Caption = "EnumWindowChild"

    Label3.Left = 3060

    Label3.Top = 3150

    Label3.Width = 2895

    Label3.Height = 240

    Label3.Caption = "Clase a la cual pertence la ventana hija"

    Label4.Left = 3060

    Label4.Top = 3375

    Label4.Width = 2715

    Label4.Height = 240

    Label4.Caption = ""

    Label4.ForeColor = &H80&

    Label4.BackColor = &HC0FFFF

    Label5.Left = 180

    Label5.Top = 3150

    Label5.Width = 2670

    Label5.Height = 465

    Label5.Caption = "Clase a la cual pertence la ventana - aplicacion"

    Label6.Left = 180

    Label6.Top = 3600

    Label6.Width = 2715

    Label6.Height = 375

    Label6.Caption = ""

    Label6.ForeColor = &H80&

    Label6.BackColor = &HC0FFFF

    List1.Left = 180

    List1.Top = 270

    List1.Width = 2670

    List1.Height = 2790

    List2.Left = 3060

    List2.Top = 270

    List2.Width = 2670

    List2.Height = 2790

    EnumWindows AddressOf EnumWindowsProc, ByVal 0&

End Sub

Private Sub List1_Click()

    Dim nombreclase As String

    Dim resultado As Long

    Dim hWnd As String

    hWnd = Mid(List1.Text, 1, InStr(1, List1.Text, " "))

    'Buscamos la clase

    nombreclase = Space(256)

    resultado = GetClassName(CLng(hWnd), nombreclase, 256)

    Label6 = Left$(nombreclase, resultado)

    'Buscamos sus ventanas hijas

    List2.Clear

    EnumChildWindows CLng(hWnd), AddressOf EnumChildProc, ByVal 0&

End Sub

Private Sub List2_Click()

    Dim nombreclase As String

    Dim resultado As Long

    Dim hWnd As String

    hWnd = Mid(List2.Text, 1, InStr(1, List2.Text, " "))

    nombreclase = Space(256)

    resultado = GetClassName(CLng(hWnd), nombreclase, 256)

    Label4 = Left$(nombreclase, resultado)

End Sub
Código del modulo:

Declare Function EnumWindows Lib "user32" (ByVal lpEnumFunc As Long, ByVal lParam As Long) As Boolean

Declare Function GetDesktopWindow Lib "user32" () As Long

Declare Function EnumChildWindows Lib "user32" (ByVal hWndParent As Long, ByVal lpEnumFunc As Long, ByVal lParam As Long) As Long

Declare Function GetWindowText Lib "user32" Alias "GetWindowTextA" (ByVal hWnd As Long, ByVal lpString As String, ByVal cch As Long) As Long

Declare Function GetWindowTextLength Lib "user32" Alias "GetWindowTextLengthA" (ByVal hWnd As Long) As Long

Declare Function GetClassName Lib "user32" Alias "GetClassNameA" (ByVal hWnd As Long, ByVal lpClassName As String, ByVal nMaxCount As Long) As Long

Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hWnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Declare Function SetWindowText Lib "user32" Alias "SetWindowTextA" (ByVal hWnd As Long, ByVal lpString As String) As Long

Declare Function SetWindowPos Lib "user32" (ByVal hWnd As Long, ByVal hWndInsertAfter As Long, ByVal X As Long, ByVal Y As Long, ByVal cx As Long, ByVal cy As Long, ByVal wFlags As Long) As Long

Public Function EnumWindowsProc(ByVal hWnd As Long, ByVal lParam As Long) As Boolean

    Dim texto As String

    texto = Space$(GetWindowTextLength(hWnd) + 1)

    GetWindowText hWnd, texto, Len(texto)

    texto = Left$(texto, Len(texto) - 1)

    If texto <> "" Then frmEnumWin.List1.AddItem Trim(Str$(hWnd) + " " + texto)

    EnumWindowsProc = True

End Function

Public Function EnumChildProc(ByVal hWnd As Long, ByVal lParam As Long) As Long

    Dim texto As String

    texto = Space$(GetWindowTextLength(hWnd) + 1)

    GetWindowText hWnd, texto, Len(texto)

    texto = Left$(texto, Len(texto) - 1)

    frmEnumWin.List2.AddItem LTrim(Str$(hWnd) + " " + texto)

    EnumChildProc = 1

End Function

Cuando arrancan el proyecto verán un formulario con 2 listbox, uno de ellos se encuentra lleno, y podrán ver que hay como dos columnas, una donde hay números y la otra donde hay un texto, ejm: “456 XXXXX”, déjenme decirle que lo que ven en ese listbox son las aplicaciones o ventanas principales abiertas en ese momento, no son todas, ya que filtre aquellas aplicaciones que no poseen “texto”, ya lo veremos cuando explique el código.

Si dan clic, en el list1, se darán cuenta que para algunas aplicaciones el list2 se llena, este list2 contiene TODAS las ventanas HIJAS que posee la aplicación principal.

¿Cómo jugar con eso?. Habrán la calculadora y pónganla en opción “ESTANDAR”. Luego cuando la tengan abierta denle al botón “R” ubicado arriba del list1, este botón lo que hace es refrescar la lista de aplicaciones abiertas.

Después de apretar “R”, se darán cuenta que la calculadora se encuentra en el list1, ella posee un número “XXXX Calculadora”, ese XXXX, (puede ser de 3 cifras, 2 cifras o mas), es el HWDN de esa ventana, ahora denle clic en el list1, a la fila que tiene la calculadora. En ese momento en el list2, se encuentran todas las ventanas hijas de la calculadora.

En el list2, debería de haber una fila que sea de esta manera “XXXX 0,” XXXX puede ser cualquier número, si se dan cuenta “0,” es el texto que se muestra en la calculadora, ¡Sorprendente!, seleccionen esa fila en el list2, cuando este seleccionada presionen el botón “Enviar Texto” que se encuentra en la aplicación.

Al presionarlo no ven nada, pero ahora minimicen la calculadora y restáurenla de nuevo. ¿Y que ven?. MAGIA!!!, mucho se deben de preguntar, ¿Dios!!!, como lo hiciste?, vamos a seguir dando sorpresas,  si dan clic a los números que no poseen texto, se darán cuenta abajo en la etiqueta la clase a la cual pertenece, y también se darán cuenta que la mayoría son de la clase “button”, es decir todos los HWND que tenga clase “button” nosotros sabemos que representa a los botones de la calculadora.

Pero si son curiosos deberían de estar preguntándose ¿Por qué no vemos como texto los números de los botones?, interesante pregunta que no he podido responder, es muy raro que no aparezca dicho texto, eso me hace pensar que los números son gráficos, pero es como ilógico, ya que en algo tan simple, ¿para que usar gráficos?, en realidad esa respuesta no la se. Pero eso no impide que sigamos jugando, ¿Cómo sabemos cual botón corresponde a los números?. Fácil, vamos a dar clic a cada fila cuyo HWND sea de la clase “button” y luego presionan el botón “ENVIAR CLICK”, tengan en cuenta que deben de ver el formulario y la calculadora al mismo tiempo, cuando den con el botón indicado verán como la calculadora escribe el número sin haber tocado directamente la calculadora.

En mi caso los números empiezan aparecer desde la fila 5 del list2. ¡sorprendente!, luego que jueguen un poco con eso, en el list1 verifiquen que este seleccionado la fila correspondiente de la calculadora y presionen el botón “DESTRUIR APLICACIÓN”, después de presionarlo verán que ya no hay calculadora, siguen viéndolo en la lista por que tienen que refrescar presionando el botón “R”.

Si siguieron todo al pie de la letra, y les funciono, se habrán dado cuenta que pudimos manipular una ventana desde otra. A continuación vamos a ver las funciones que permitieron realizar esa magia.

EnumWindows

Api: Declare Function EnumWindows Lib "user32" (ByVal lpEnumFunc As Long, ByVal lParam As Long) As Boolean

Esta función es similar “en como funciona” a la función que se encargaba de enumerar las fuentes, lo que se le pasa a esta función es la dirección de otra función que cumple ciertas características, dicha función va ser invocada por cada ventana que este abierta.

En conclusión, esta función enumera las ventanas principales abiertas del sistema, como primer parámetro le pasamos la dirección de memoria de la función que procesara la información de cada ventana, y como segundo parámetro le pasamos alguna información adicional que nosotros creamos conveniente.

La función que se le pasa en el primer parámetro debe de cumplir los siguientes requisitos o parámetros:

Public Function EnumWindowsProc(ByVal hWnd As Long, ByVal lParam As Long) As Boolean

Esta función solo tiene 2 parámetros el primero, es el “hwnd” de la ventana abierta y como segundo parámetro es la información que mandamos como segundo parámetro de la función EnumWindows.

Lo que quiero que vean es que, lo que hace el programa es entrar a esta función tantas veces como ventana abiertas existan. Y cada ves que entra es con un “hwnd” distinto.

EnumChildWindows

Api: Declare Function EnumChildWindows Lib "user32" (ByVal hWndParent As Long, ByVal lpEnumFunc As Long, ByVal lParam As Long) As Long

Esta función es la encargada de enumerar todas las ventanas hijas de una ventana padre.

Casi igual a la anterior, con la diferencia, que como primer parámetro tenemos un “hwnd”,  el cual es el padre del cual queremos ver que ventana hijas posee, el segundo parámetro es la dirección de la función a la cual se le pasa la información de la ventana hija y como tercer parámetro tenemos la información adicional que puede ser enviada a la función.

Esa función debe de cumplir la siguiente estructura:

Public Function EnumChildProc(ByVal hWnd As Long, ByVal lParam As Long) As Long

Como pueden ver es exactamente igual a la función anterior, con la diferencia que esta función se va a ejecutar tantas veces como ventanas hijas posea el padre que fue pasado en el parámetro “hWndParent”

¿Qué hacemos dentro de esas funciones?. ¡Lo que quieran!, ya que poseen el HWND, número clave al momento de manipular cualquier ventana, en nuestro caso, lo que hice fue tomar el texto de la ventana y para ello use 2 funciones claves.

GetWindowTextLength

Api: Declare Function GetWindowTextLength Lib "user32" Alias "GetWindowTextLengthA" (ByVal hWnd As Long) As Long

Esta función lo que hace es retornar la longitud del texto que tiene la ventana ¿qué texto?. Se recuerdan en la función CreateWindowEx, el parámetro “LpWindowName” ese es el texto al cual me refiero, o mas fácil aun, en un formulario es el titulo que esta en la barra del caption, en un botón sería el texto que muestra, y así dependiendo del control  o tipo de ventana que sea.

Esta función la utilizamos para reservar el espacio en memoria para dicho texto, ya que la función que toma el texto es la siguiente:

GetWindowText

Api: Declare Function GetWindowText Lib "user32" Alias "GetWindowTextA" (ByVal hWnd As Long, ByVal lpString As String, ByVal cch As Long) As Long

Esta función es la encargada de tomar el texto de una ventana, como primer parámetro le pasamos el “hwnd” de la ventana a la cual queremos tomar el texto, como segundo parámetro, pasamos la variable que almacenara dicho texto, y como tercer parámetro le pasamos la cantidad de caracteres que queremos recuperar.

¿Cómo usamos todo lo mencionado en nuestro proyecto?.

En el LOAD del formulario tenemos esta línea:

EnumWindows AddressOf EnumWindowsProc, ByVal 0&

Con esta línea activamos la función EnumWindowsProc la cual se ejecutara tantas veces como ventanas principales abiertas existan. Dicha función esta formada por el siguiente bloque de código.

Public Function EnumWindowsProc(ByVal hWnd As Long, ByVal lParam As Long) As Boolean

    Dim texto As String ‘ Declaramos la variable de tipo texto que almacenara el nombre de la ventana

    texto = Space$(GetWindowTextLength(hWnd) + 1) ‘ Reservamos en memoria el espacio para dicho nombre, en pocas palabras, llenamos con espacio en blanco la variable “texto”.
    GetWindowText hWnd, texto, Len(texto) ‘ Tomamos el texto de la ventana.
    texto = Left$(texto, Len(texto) - 1) ‘Quitamos el último carácter, que es un espacio en blanco.
    If texto <> "" Then frmEnumWin.List1.AddItem Trim(Str$(hWnd) + " " + texto) ‘ Si el nombre es distinto a “” entonces lo añado al listado.
    EnumWindowsProc = True ‘Retorno TRUE para que la función siga buscando ventanas, ya que en caso de poner FALSE, pues la enumeración de las ventanas se para.
End Function
Pueden ver que hago If texto <> "" Then , si ustedes quitan este IF verán TODAS las ventanas abiertas con nombre y sin nombres. Particularmente una ventana sin nombre, es muy difícil saber que es.

Pueden preguntarse ¿Por qué hay nombre tan raros?. Y ¿Por qué hay mas ventanas abiertas de las que puedo ver?. Con esto se dan cuenta de los procesos que Windows maneja internamente, hay programas no activados por nosotros sino por el sistema operativo, algunos programas pueden ser pequeños o carecer de ventanas hijas. Estas aplicaciones no son controladas por nosotros sino por Windows, por eso  es que tienen que tener cuidado con el botón “Destruir Aplicación”, úsenlo siempre y cuando sepan que aplicación van a destruir, no destruyan cualquier aplicación ya que puede producir que Windows se quede colgado.

La explicación de arriba vale perfectamente para la función EnumChildProc con la diferencia que esta llena el list2, cada vez que damos clic en una fila del list1.

Cada vez que daban clic a uno de los 2 listbox, se daban cuenta que en una etiqueta en la parte de abajo, llenábamos dicha etiqueta con el nombre de la clase a la cual pertenece la aplicación, a continuación vamos a explicar la función encargada de retornarnos el nombre de dicha clase.

GetClassName

Api: Declare Function GetClassName Lib "user32" Alias "GetClassNameA" (ByVal hWnd As Long, ByVal lpClassName As String, ByVal nMaxCount As Long) As Long

Con esta función tomamos el nombre de la clase de cierta ventana. Como primer parámetro le pasamos el “hwnd” de la ventana, como segundo parámetro le pasamos la variable de tipo String que almacenara dicho nombre, y como tercer parámetro le pasamos la cantidad de caracteres que queremos retornar.

El valor que retorna dicha función es la longitud del nombre de la clase.

En el ejemplo lo usamos en List1_Click de la siguiente manera:

    hWnd = Mid(List1.Text, 1, InStr(1, List1.Text, " ")) Con este codigo tomamos el HWND del listbox, recuerden que el listbox tiene “XXXX NOMBRE DE LA VENTANA” esta línea de codigo toma el XXXX que es el hwnd de la ventana

    'Buscamos la clase

    nombreclase = Space(256) Reservamos 256 espacios
    resultado = GetClassName(CLng(hWnd), nombreclase, 256) Llamamos a la función pasándole la variable “nombreclase” variable que almacenara dicho nombre y en la variable “resultado” almacenamos la longitud de la cadena
    Label6 = Left$(nombreclase, resultado) Quitamos los espacios en blanco e imprimimos el nombre en la etiqueta.
La explicación vale para List2_Click

GetDesktopWindow

Api: Declare Function GetDesktopWindow Lib "user32" () As Long

Esta función retorna el “hwnd” del escritorio de Windows, esta función lo usamos cuando le damos clic al botón "Enumerar los Child del Destkop de Windows". Que tiene el siguiente codigo:

EnumChildWindows GetDesktopWindow, AddressOf EnumChildProc, ByVal 0&

Al apretar clic en este botón llenamos el list2 con TODAS las ventanas abiertas,  como el padre es el escritorio de Windows, en el list2 verán TODO!, cuando me refiero a todo, hablo de ventanas principales, y sus respectivas ventanas hijas, por esa razón algunas filas del list2 se repiten en el list1.

SetWindowText

Declare Function SetWindowText Lib "user32" Alias "SetWindowTextA" (ByVal hWnd As Long, ByVal lpString As String) As Long

Función utilizada para asignar un texto en una ventana, como primer parámetro le pasamos el “hwnd” de la ventana, y como segundo parámetro le pasamos el texto que queremos mostrar.

El uso de esta función, creo que es simple y se encuentra en el evento Command3_Click
Mensajes

¿Cuándo estábamos con la calculadora, como enviábamos los mensajes?. Usando la función SendMessage, pero vamos a ver el codigo:

Private Sub Command2_Click()

    Dim hWnd As String

    hWnd = Mid(List1.Text, 1, InStr(1, List1.Text, " ") - 1) ‘ Tomamos el “hwnd” de la ventana

    SendMessage CLng(hWnd), WM_CLOSE, 0, 0 ‘Enviamos el mensaje WM_CLOSE, mensaje encargado de cerrar una aplicación
End Sub

En el otro botón:

Private Sub Command4_Click()

    Dim hWnd As String

    hWnd = Mid(List2.Text, 1, InStr(1, List2.Text, " ") - 1)

    SendMessage CLng(hWnd), BM_CLICK, 0, 0

End Sub

Noten que enviamos el mensaje BM_CLICK, ese es un mensaje exclusivo de los botones, cuando un botón es pulsado Windows le manda el mensaje BM_CLICK para indicarle al botón que se le ha presionado. Este mensaje no es un mensaje para las ventanas en general, es solo usado para los controles de la clase “button” y que deriven de esa clase. Ya que el botón de VB no es de clase “button”, pero acepta el mensaje “BM_CLICK” 

La función SetWindowPos no lo explico aquí ya que se explica en la sección que viene a continuación.

BUSCANDO Y POSICIONANDO VENTANAS

Para este proyecto es necesario: 2 Botones, 1 TextBox y 1 Label. Luego pegan este codigo en el formulario:

Private Declare Function SetWindowPos Lib "user32" (ByVal hwnd As Long, ByVal hWndInsertAfter As Long, ByVal x As Long, ByVal y As Long, ByVal cx As Long, ByVal cy As Long, ByVal wFlags As Long) As Long

Private Declare Function FindWindow Lib "user32" Alias "FindWindowA" (ByVal lpClassName As String, ByVal lpWindowName As String) As Long

Private Declare Function FindWindowEx Lib "user32" Alias "FindWindowExA" (ByVal hWnd1 As Long, ByVal hWnd2 As Long, ByVal lpsz1 As String, ByVal lpsz2 As String) As Long

Private Const HWND_BOTTOM = 1

Private Const HWND_NOTOPMOST = -2

Private Const HWND_TOP = 0

Private Const HWND_TOPMOST = -1

Private Const SWP_ASYNCWINDOWPOS = &H4000

Private Const SWP_DEFERERASE = &H2000

Private Const SWP_FRAMECHANGED = &H20

Private Const SWP_DRAWFRAME = SWP_FRAMECHANGED

Private Const SWP_HIDEWINDOW = &H80

Private Const SWP_NOACTIVATE = &H10

Private Const SWP_NOCOPYBITS = &H100

Private Const SWP_NOMOVE = &H2

Private Const SWP_NOOWNERZORDER = &H200

Private Const SWP_NOREDRAW = &H8

Private Const SWP_NOREPOSITION = SWP_NOOWNERZORDER

Private Const SWP_NOSENDCHANGING = &H400

Private Const SWP_NOSIZE = &H1

Private Const SWP_NOZORDER = &H4

Private Const SWP_SHOWWINDOW = &H40

Private Sub Command1_Click()

    If Command1.Caption = "Siempre en el top" Then

       SetWindowPos Me.hwnd, HWND_TOPMOST, 0, 0, 0, 0, SWP_SHOWWINDOW Or SWP_NOSIZE Or SWP_NOMOVE

       Command1.Caption = "No siempre en el top"

    Else

       SetWindowPos Me.hwnd, HWND_NOTOPMOST, 0, 0, 0, 0, SWP_SHOWWINDOW Or SWP_NOSIZE Or SWP_NOMOVE

       Command1.Caption = "Siempre en el top"

    End If

End Sub

Private Sub Command2_Click()

    Dim resultado As Long

    resultado = FindWindow(vbNullString, Trim(Text1.Text))

    If resultado = 0 Then

        Label1 = "Ventana no existente"

    Else

        Label1 = "La ventana '" & Text1.Text & "' existe y su HWND es = " & resultado

    End If

End Sub

Private Sub Form_Load()

    Command1.Top = 90

    Command1.Left = 90

    Command1.Width = 2985

    Command1.Height = 510

    Command1.Caption = "Siempre en el top"

    Command2.Top = 1125

    Command2.Left = 90

    Command2.Width = 2895

    Command2.Height = 510

    Command2.Caption = "Buscar Ventana"

    Label1.Top = 1710

    Label1.Left = 90

    Label1.Width = 2895

    Label1.Height = 600

    Label1.ForeColor = &H80&

    Label1.BackColor = &HC0FFFF

    Label1.Caption = ""

    Text1.Top = 765

    Text1.Left = 90

    Text1.Width = 2895

    Text1.Height = 285

    Text1.Text = ""

    Me.Width = 3255

    Me.Height = 2880

End Sub
Para explicar lo que hace el programa iremos directo a las funciones.

SetWindowPos

Api: Declare Function SetWindowPos Lib "user32" (ByVal hwnd As Long, ByVal hWndInsertAfter As Long, ByVal x As Long, ByVal y As Long, ByVal cx As Long, ByVal cy As Long, ByVal wFlags As Long) As Long

Esta función lo que hace es cambiar el posicionamiento de la ventana y su tamaño. Como primer parámetro tenemos el “hwnd” que queremos alterar. 

El segundo parámetro determina en que posición vamos a colocar nuestra ventana, este parámetro puede tomar los siguientes valores:

Private Const HWND_BOTTOM = 1 : Envía la ventana al fondo, es decir si hay 10 ventanas abiertas y la ventana que queremos alterar esta en el inicio, esta es enviada al final de todas las demás ventanas.

Private Const HWND_NOTOPMOST = -2 : Coloca la ventana por encima de todas las ventanas no-topmmost. Pero queda detrás de aquellas que estén en el topmost

Private Const HWND_TOP = 0 : Coloca la ventana en el TOPE es decir de primera.

Private Const HWND_TOPMOST = -1 : Coloca la ventana en el TOPE, y se mantiene en esa posición inclusive si la ventana se encuentra desactivada.

Esta ultima opción es la utilizada, cuando usamos formularios de presentación en el inicio de nuestro programa, como por ejemplo, la ventana de inicio de Word, que se mantiene en el inicio hasta que la aplicación se encuentra cargada completamente, inclusive VB también lo utiliza.

Para ver tal efecto aprieten el botón  “Siempre en el Top”, y verán que no importa que ventana intente superponer, siempre nuestro formulario estará de primero, para regresarla a la normalidad vuelvan a presionar el botón.

Parámetros x, y, cx, cy : los dos primeros, determinan las nuevas coordenadas x e y donde será colocada la ventana y cx, cy, determinan el nuevo alto y ancho de la ventana.

Tenemos como último parámetro wFlags, que puede tomar los siguientes valores (dichos valores pueden ser combinados con el operador OR):

Private Const SWP_ASYNCWINDOWPOS = &H4000 

Private Const SWP_DEFERERASE = &H2000 : Previene la generación del mensaje WM_SYNCPAINT

Private Const SWP_FRAMECHANGED = &H20 : Dibuja el marco alrededor de la ventana

Private Const SWP_DRAWFRAME = SWP_FRAMECHANGED 

Private Const SWP_HIDEWINDOW = &H80 : Esconde la ventana

Private Const SWP_NOACTIVATE = &H10 : No activa la ventana, no importa si colocaron la ventana en el inicio, no se activa.

Private Const SWP_NOCOPYBITS = &H100 

Private Const SWP_NOMOVE = &H2 : Retiene la posicion actual de la ventana, es decir, hace caso omiso a los parámetro X y Y de la función.

Private Const SWP_NOOWNERZORDER = &H200 : No se realiza ningún cambio de posicionamiento a la ventana padre.

Private Const SWP_NOREDRAW = &H8 : No repinta la ventana.

Private Const SWP_NOREPOSITION = SWP_NOOWNERZORDER

Private Const SWP_NOSENDCHANGING = &H400 : previne a la ventana para recibir el mensaje WM_WINDOWPOSCHANGING

Private Const SWP_NOSIZE = &H1 : Mantiene el ancho y la altura actual, es decir, hace caso omiso a los parámetros cx e cy.

Private Const SWP_NOZORDER = &H4 : Mantiene la posicion actual de la ventana, es decir, hace caso omiso al valor pasado en el parámetro hWndInsertAfter.

Private Const SWP_SHOWWINDOW = &H40 : Despliega la ventana.

FindWindow

Api: Declare Function FindWindow Lib "user32" Alias "FindWindowA" (ByVal lpClassName As String, ByVal lpWindowName As String) As Long

Función que se encarga de buscar a una ventana, hay dos formas de buscar una ventana o por su clase, o por su nombre, en caso de ser por su clase, deberán de colocar el nombre de la clase en el primer parámetro, y poner “vbnullstring” en el segundo, en caso de buscarla por nombre, pues colocan el nombre en el segundo parámetro y “vbnullstring” en el primero. 

Si la búsqueda es exitosa entonces retorna el “hwnd” de dicha ventana, en caso de fallar la función retornara  0.

¿Para que buscar una ventana?. 

Una vez, tenia que hacer un programa para un Cyber Café (centro de internet), el cliente quería que cuando la computadora estuviera inactiva, es decir, que no hubiese nadie en ella, en caso de que una persona mal intencionada se sentara y tratara de abrir el IE no pudiera, y no solo el IE sino cualquier programa de Office o los programas básicos de una computadora. Para solucionar ese problema hice un programa que se mantenía en memoria y buscaba los programas. Si una persona sin autorización habría el IE el programa mediante la función FindWindow encontraba que la aplicación se encontraba abierta y la cerraba automáticamente. Esta técnica es un poco obsoleta, pero estamos hablando de hace un año y medio. Con esto les quiero decir, que si hay casos posibles para usar esta función.

¿Es mejor usar clase o el nombre de la ventana?.

Resulta que hay ventanas cuyo titulo cambia dependiendo de su contenido, ejemplo fácil es Microsoft Word, el titulo de la ventana depende del archivo que este abierto. ¿Cómo saber si Word esta abierto?. Por el titulo de la ventana sería imposible, entonces recurrimos a la clase, inclusive cuando salen nuevas versiones, los programas suelen mantener el mismo nombre de clase, con esto pudieran hacer un programa que verificara si una aplicación esta abierta sin importar la versión que se abre. ¿Cómo sabemos el nombre de la clase?. Ya les di el truco en el proyecto anterior a este. Inclusive en el programa del proyecto anterior, debajo del list1, hay una etiqueta que muestra la clase de la aplicación seleccionada.

¿Pero, tu nos dijiste que pueden haber ventanas con el mismo nombre de la clase?.

Cierto, y en caso de que haya mas de una ventana abierta con la misma clase, la función FindWindow retorna la ventana que este mas arriba, es decir, la que este mas hacia el TOPE o TOP. Pero vamos a estar claro, las nombres de las clases no suelen ser fáciles de copiar, la probabilidades que ustedes creen una aplicación con el mismo nombre de la clase de Word, por ejemplo, sería casi imposible, y también es casi imposible encontrar 2 aplicaciones comerciales a nivel mundial que posean el mismo nombre de clase. OJO!! En este me refiero a la ventana PRINCIPAL. Es evidente que 2 aplicaciones pueden compartir ventanas hijas de la misma clase, pero la clase de la ventana principal suele ser muy difícil que se repita en software comerciales. Es posible que ustedes desarrollen clases propias para uso interno y se de el caso, pero es muy raro verlo.

En el programa lo que tienen que hacer, es poner el nombre de una ventana en el Text1 y si la encuentra muestra en el Label1 el nombre de la clase, para hacerlo sencillo, habrán la calculadora de Windows, cuyo titulo es “Calculadora” pónganlo en el text1 luego presionen el botón de buscar y el programa mostrara la clase correspondiente a esa ventana.

FindWindowEx

Api: Declare Function FindWindowEx Lib "user32" Alias "FindWindowExA" (ByVal hWnd1 As Long, ByVal hWnd2 As Long, ByVal lpsz1 As String, ByVal lpsz2 As String) As Long

La función de arriba solo buscaba ventanas principales o “main window”, esta función hace exactamente lo mismo, pero busca ventanas hijas dentro de una ventana padre.

Como primer parámetro le pasamos el “hwnd” de la ventana padre. Como segundo parámetro le pasamos el “hwnd” de la ventana hija de donde se empezara a buscar, usualmente se le coloca 0, para decirle a la función que busque en todas las ventanas hijas existentes.

El tercer parámetro equivale al nombre de la clase y el cuarto parámetro equivale al nombre de la ventana, y se trabaja de la misma manera que FindWindow.

Si la búsqueda es exitosa entonces retorna el “hwnd” de dicha ventana, en caso de fallar la función retornara  0.

Con esta función si puede haber la posibilidad de buscar en ventanas que puedan tener el mismo nombre de la clase e inclusive el mismo nombre de la ventana, y al igual que Findwindow, la función retornara el “hwnd” de la ventana que esta mas arriba o hacia el TOP.

ESTILOS Y PARIENTES

Para este proyecto es necesario 2 formularios, el formulario de arranque tendrá el nombre “frmWindowsstyle” y el segundo formulario será “frmHija”, en el formulario “frmWindowsstyle”, colocaran 3 botones y una etiqueta Label. Luego le colocan este código:

Private Type POINTAPI

    X As Long

    Y As Long

End Type

Private Declare Function GetWindowLong Lib "user32" Alias "GetWindowLongA" (ByVal hWnd As Long, ByVal nIndex As Long) As Long

Private Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hWnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Private Declare Function ChildWindowFromPointEx Lib "user32" (ByVal hWnd As Long, ByVal xPoint As Long, ByVal yPoint As Long, ByVal un As Long) As Long

Private Declare Function SetParent Lib "user32" (ByVal hWndChild As Long, ByVal hWndNewParent As Long) As Long

Private Declare Function SetWindowPos Lib "user32" (ByVal hWnd As Long, ByVal hWndInsertAfter As Long, ByVal X As Long, ByVal Y As Long, ByVal cx As Long, ByVal cy As Long, ByVal wFlags As Long) As Long

Private Declare Function ChildWindowFromPoint Lib "user32" (ByVal hWnd As Long, ByVal xPoint As Long, ByVal yPoint As Long) As Long

Const CWP_ALL = &H0

Const WS_MAXIMIZE = &H1000000

Const WS_CAPTION = &HC00000                  '  WS_BORDER Or WS_DLGFRAME

Const GWL_STYLE = (-16)

Const SWP_SHOWWINDOW = &H40

Const HWND_TOP = 0

Dim punto As POINTAPI

Private Sub Command1_Click()

    Me.Hide

    r = SetWindowLong(Me.hWnd, GWL_STYLE, WS_CAPTION Or WS_MAXIMIZE)

    SetWindowPos Me.hWnd, HWND_TOP, 10, 10, 300, 300, SWP_SHOWWINDOW

    Me.Refresh

End Sub

Private Sub Command2_Click()

    SetParent frmHija.hWnd, Me.hWnd

End Sub

Private Sub Command3_Click()

    Dim hwndresultado As Long

    hwndresultado = ChildWindowFromPointEx(Me.hWnd, 100 \ 15, 140 \ 15, CWP_ALL)

    Select Case hwndresultado

        Case Me.hWnd

            Label1 = "Estas en el formulario"

        Case command1.hWnd

            Label1 = "Estas en el botón 1"

        Case Command2.hWnd

            Label1 = "Estas en el botón 2"

        Case frmHija.hWnd

            Label1 = "Estas en el formulario hijo"

    End Select

End Sub

Private Sub Form_Load()

    command1.Left = 90

    command1.Top = 135

    command1.Width = 2130

    command1.Height = 510

    command1.Caption = "Cambiar Estilo"

    Command2.Left = 135

    Command2.Top = 720

    Command2.Width = 2085

    Command2.Height = 510

    Command2.Caption = "Cambiar Padre"

    Command3.Left = 135

    Command3.Top = 1305

    Command3.Width = 2055

    Command3.Height = 510

    Command3.Caption = "Detectar"

    Label1.Left = 135

    Label1.Top = 1980

    Label1.Width = 2085

    Label1.Height = 510

    Label1.Caption = ""

    frmHija.Show

End Sub

Private Sub Form_Unload(Cancel As Integer)

    Unload frmHija

End Sub
Al arrancar la aplicación, verán que los dos formularios arrancan, esto sucede, por que el formulario de arranque llama al formulario “frmHija”, estos 2 formularios no tienen relación alguna, el padre de ambos formularios es el escritorio de Windows, pero ustedes dirán ¿Pero cuando cerré la aplicación los 2 formularios se cerraron?. Esto es por que en el UNLOAD del formulario de arranque, yo descargo el formulario “frmHija”.

En este proyecto trabajamos el cambio de estilos y el cambio de padre, presionen el botón “Cambiar Padre”, verán como el formulario “frmhija” es incrustado dentro del formulario de arranque, a partir de este momento el padre de “frmhija” es el formulario “frmWindowsstyle”.

Ahora presionen el botón “Cambiar Estilo”, notaran que el formulario” frmWindowsstyle”, cambia de estilo, ya que los botones del área no cliente y el menú del sistema se desaparecen. ¿Cómo lo hicimos? Veamos las funciones:

SetWindowLong

Api: Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hWnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Esta función es utilizada para cambiar los atributos de una ventana.

Como primer parámetro se le pasa el”hwnd” de la ventana cuyos atributos queremos cambiar. Como segundo parámetro pasamos una constante que identifica lo que queremos cambiar, esta puede tomar los siguientes valores:

Public Const GWL_EXSTYLE = (-20) : Con este parámetro le indicamos a la función que lo que queremos cambiar son los estilos “extras”

Public Const GWL_HINSTANCE = (-6) : Con este indicamos que cambiamos la instancia de la ventana

Public Const GWL_ID = (-12) : Le ponemos un nuevo identificador a la ventana

Public Const GWL_STYLE = (-16) : Le cambiamos los estilos a la ventana

Public Const GWL_WNDPROC = (-4) : Le cambiamos el gestor de mensaje a la ventana

Public Const GWL_USERDATA = (-21)

En nuestro proyecto usamos la constante GWL_STYLE, es decir, cambiamos el estilo de la ventana.

Como tercer parámetro pasamos el nuevo valor a cambiar. Veamos el ejemplo:

Private Sub Command1_Click()

    Me.Hide  ‘Escondo la ventana para evitar interferencia

    r = SetWindowLong(Me.hWnd, GWL_STYLE, WS_CAPTION Or WS_MAXIMIZE) ‘ Vean que le estoy asignando 2 nuevos estilos, es decir, los viejos se eliminan y se sustituyen por los estilos coloreados en azul

    SetWindowPos Me.hWnd, HWND_TOP, 10, 10, 300, 300, SWP_SHOWWINDOW ‘ repinto la ventana para visualizar los cambios.

    Me.Refresh ‘ Refresco el formulario

End Sub

Si la función se ejecuta exitosamente, el valor que regresa, es el valor anterior que poseía la ventana, es decir, en nuestro ejemplo, la variable “r” tiene los estilos que tenia la ventana anteriormente, si cambiamos la instancia, “r” tendría la instancia anterior de la ventana, y así con los demás cambios.

Mas adelante vamos a ver como cambiar el gestor de mensaje.

GetWindowLong

Api: Declare Function GetWindowLong Lib "user32" Alias "GetWindowLongA" (ByVal hWnd As Long, ByVal nIndex As Long) As Long

Esta función aunque no  es utilizada, es útil cuando queremos tomar información de la ventana.  Como primer parámetro le pasamos el “hwnd” de la ventana, de la cual extraeremos dicha información, y como segundo parámetro especificamos por una constante que información queremos tomar. Estas constante son iguales a las constantes explicadas en SetWindowLong, pero esta función toma una constante mas, que es:

Public Const GWL_HWNDPARENT = (-8) : Este parámetro permite obtener el “hwnd” de la ventana padre.

Un ejemplo sería algo así:

r = GetWindowLong(Me.hWnd, GWL_STYLE)

En este caso “r” tendrá el valor de los estilos de la ventana, y el valor “r” dependerá de la constante que haya sido pasado en “nIndex”.

Noten que con GetWindowLong podemos tomar el “hwnd” de la ventana padre con la constante GWL_HWNDPARENT, pero SetWindowLong no sirve para cambiar la ventana padre, para hacer eso contamos con la siguiente función:

SetParent

Api: Declare Function SetParent Lib "user32" (ByVal hWndChild As Long, ByVal hWndNewParent As Long) As Long

Función sencilla, como primer parámetro pasamos el “hwnd” que queremos que sea “hijo”, y como segundo parámetro pasamos el “hwnd” que será padre de la ventana identificada en el primer parámetro.

En nuestro proyecto usamos la función de la siguiente manera:

    SetParent frmHija.hWnd, Me.hWnd

Creo que es sencillo ver lo que hace.

ChildWindowFromPointEx

Api: Declare Function ChildWindowFromPointEx Lib "user32" (ByVal hWnd As Long, ByVal xPoint As Long, ByVal yPoint As Long, ByVal un As Long) As Long

Esta función es utilizada para saber si hay una ventana hija en una ventana padre y en una coordenada especifica.

Como primer parámetro pasamos el “hwnd” de la ventana padre, los parámetros “xPoint” y “yPoint”, son las coordenadas que queremos evaluar, el cuarto parámetro “un” es como el filtro del tipo de ventanas a tomar en cuenta, este parámetro puede tomar los siguientes valores:

Public Const CWP_ALL = &H0 : Toma en cuenta todas las ventanas

Public Const CWP_SKIPDISABLED = &H2 : Toma en cuenta todas las ventanas excepto las que están desactivadas.

Public Const CWP_SKIPINVISIBLE = &H1 : Toma en cuenta todas las ventanas excepto las que están invisibles.

Public Const CWP_SKIPTRANSPARENT = &H4 : Toma en cuenta todas las ventanas excepto las que están transparente.

En caso de encontrar alguna ventana hija en dicha coordenada, la función retorna el “hwnd” de dicha ventana.

¿Qué significa esto?. Veamos el ejemplo:

hwndresultado = ChildWindowFromPointEx(Me.hWnd, 100 \ 15, 140 \ 15, CWP_ALL)

Intencionalmente coloque las coordenadas donde esta el botón “command1”, es por esa razón que el label1 muestra "Estas en el botón 1", si yo pusiera la siguiente línea:

hwndresultado = ChildWindowFromPointEx(Me.hWnd, 100 \ 15, 140 \ 15, CWP_SKIPINVISIBLE)

Y ponen el botón “command1” invisible, es decir, la propiedad “visible” en false, el programa mostrara "Estas en el formulario", ya que el programa esta saltando todas las ventanas invisibles, si lo dejan como esta y vuelven a colocar CWP_ALL, el programa mostrara de nuevo "Estas en el botón 1", sin importar que el botón este visible o no.

SUBCLASSING

Un día una persona en el foro de VB en la web del programador, puso una pregunta, donde preguntaba si existía la posibilidad de que cuando el mouse se moviera por encima de un control ComboBox, este desplegara el listado que él contiene.

Me sorprendí por la pregunta ya que parecía muy sencillo, hasta que me di cuenta que el control ComboBox no poseía el evento MouseMove como los demás controles, pero la persona en su pregunta daba como IMPOSIBLE, encontrar dicha solución.

En ese momento, le escribí diciéndole, antes que nada, “que en la programación no hay nada imposible pero si difícil”, y luego le mostré el código que les presento a continuación.

Lo único que necesitan es un ComboBox en el formulario y un modulo.

En el modulo ponen este codigo:

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Public Const CB_SHOWDROPDOWN = &H14F

Public Const GWL_WNDPROC = (-4)

Const WM_MOUSEMOVE = &H200

Const WM_MOUSELEAVE = &H2A3

Dim ViejoGestor As Long

Public entro As Boolean

Public Sub nuevoGestor(hwnd As Long)

    ViejoGestor = SetWindowLong(hwnd, GWL_WNDPROC, AddressOf GestorMensaje)

End Sub

Public Function GestorMensaje(ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    If mensaje = WM_MOUSEMOVE Then

        SendMessage hwnd, CB_SHOWDROPDOWN, True, 0

    End If

    GestorMensaje = CallWindowProc(ViejoGestor, hwnd, mensaje, wParam, lParam)

End Function

En el formulario:

Private Sub Form_Load()

    Combo1.AddItem "Hola"

    Combo1.AddItem "Hola 1"

    Combo1.AddItem "Hola 2"

    entro = False

    nuevoGestor Combo1.hwnd

End Sub

Si arrancan el programa, sorprendentemente cuando pasen el mouse por encima del control este habilita o muestra el listado que contiene. Lo que la persona considero como IMPOSIBLE si hizo realidad. Y espero que haya aprendido la lección.

Lo que hicimos fue cambiar nuestro “oído” o gestor de mensajes, en este caso el “oído” del control ComboBox, a este truquito se le llama SubClassing, es decir, el cambiar el gestor de mensaje de un “hwnd” cualquiera se le denomina con este nombre. Recuerden que algunos de los proyecto del Capítulo  sobre GDI usamos Subclassing pero no lo mencione directamente. Inclusive las API que intervenían en dicho proceso no las explique, ya que este es el Capítulo  que trata de ellas.

En este proyecto solo hay una función nueva que explicar. Pero vamos analizar el codigo que usamos:

ViejoGestor = SetWindowLong(hwnd, GWL_WNDPROC, AddressOf GestorMensaje)

Ya vimos que hace la función SetWindowLong, en este caso usamos la constante GWL_WNDPROC  para indicar que vamos a cambiar el gestor de mensaje. Pasando como tercer parámetro la ubicación en la memoria de la nueva función que gestionara nuestros mensajes. La variable ViejoGestor contiene la dirección de memoria de la función original que VB le pone el control ComboBox.

En el nuevo gestor de mensajes llamado “GestorMensaje”, capturamos el evento WM_MOUSEMOVE y si la variable “mensaje” es igual a ese evento ejecutamos esta línea:

SendMessage hwnd, CB_SHOWDROPDOWN, True, 0

Lo que hacemos aquí es enviarle un mensaje a nuestro comboBox indicándole que abra o despliegue su listado, este mensaje, fue visto en el Capítulo  II. Luego, como no vamos a capturar todos los mensajes del comboBox, sino nada mas el WM_MOUSEMOVE, usamos esta función:

GestorMensaje = CallWindowProc(ViejoGestor, hwnd, mensaje, wParam, lParam)

Aquí lo que estamos haciendo es llevando el mensaje al gestor de mensajes que originalmente tenia nuestro ComboBox.

CallWindowProc

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Ya sabemos lo que hace la función, llama o envía un mensaje a un gestor de mensajes que se encuentre en memoria, este gestor es pasado en el primer parámetro “lpPrevWndFunc” de la función, los otros parámetro son conocidos por ustedes.

“HABLANDO” ENTRE VENTANAS

En este proyecto vamos a ver una función que nos permite crear nuestros propios mensajes, que pueden ser utilizados para comunicar 2 aplicaciones. NOTA: estoy hablando de una comunicación local y no! en red.

Para este proyecto se necesitan un formulario, un modulo y un boton adentro del formulario.

En el formulario colocan este codigo:

Private Sub Command1_Click()

    SendMessage FindWindow(vbNullString, "Formulario 2"), MS_PERSONAL, 0, 0

End Sub

Private Sub Form_Load()

    MS_PERSONAL = RegisterWindowMessage("MS_PERSONAL")

    nuevoGestor Me.hwnd

    Me.Caption = "Formulario 1"

    Command1.Caption = "Mandar Mensaje"

End Sub

En el modulo:

Declare Function FindWindow Lib "user32" Alias "FindWindowA" (ByVal lpClassName As String, ByVal lpWindowName As String) As Long

Declare Function RegisterWindowMessage Lib "user32" Alias "RegisterWindowMessageA" (ByVal lpString As String) As Long

Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Public Const GWL_WNDPROC = (-4)

Public MS_PERSONAL As Long

Dim ViejoGestor As Long

Public entro As Boolean

Public Sub nuevoGestor(hwnd As Long)

    ViejoGestor = SetWindowLong(hwnd, GWL_WNDPROC, AddressOf GestorMensaje)

End Sub

Public Function GestorMensaje(ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    Select Case mensaje

    Case MS_PERSONAL

        MsgBox "Se he recibido un mensaje en formulario 1"

    End Select

    GestorMensaje = CallWindowProc(ViejoGestor, hwnd, mensaje, wParam, lParam)

End Function

Luego generan un archivo .exe, es decir, compilan este codigo. Pero necesitamos otro .exe, pero al codigo de arriba le van a cambiar lo siguiente:

Me.Caption = "Formulario 1" = Me.Caption = "Formulario 2"

MsgBox "Se he recibido un mensaje en formulario 1" = MsgBox "Se he recibido un mensaje en formulario 2"

SendMessage FindWindow(vbNullString, "Formulario 2"), MS_PERSONAL, 0, 0 = SendMessage FindWindow(vbNullString, "Formulario 1"), MS_PERSONAL, 0, 0

Es decir, las lineas del codigo que estan en negro, las cambian por la que estan en azul, y luego generan otro archivo .exe, con diferente nombre.

Cuando tengan los 2 archivos .exe creados, procedan a ejecutarlos, cuando presionen clic en el boton veran que le envian un mensaje a la otra aplicación, esto lo pueden ver, por el contenido del “msgbox”. ¿Cómo hicimos eso?, algunos prodrian decir, que dicho efecto lo hicimos con la calculadora en uno de los codigos de arriba ¿Cuál es la diferencia?. La diferencia esta, es que estamos comunicando 2 aplicaciónes con un mensaje privado o personalizado, en los ejemplos de arriba, usabamos mensajes, que son globales para todas las ventanas, pero en ocasiones puede ser importante comunicar 2 aplicaciones para que interactuan entre si.

RegisterWindowMessage

Api: Declare Function RegisterWindowMessage Lib "user32" Alias "RegisterWindowMessageA" (ByVal lpString As String) As Long

Esta función se encarga de registrar un mensaje, como primer parametro se le pasa una cadena de caracteres, si logra registrar dicho mensaje la función retorna el ID del mensaje. Si otra aplicación trata de registrar un mensaje ya registrado, este no se registra de nuevo, lo que hace la función es retornar el ID del mensaje.

En nuestro proyecto lo usamos de esta manera:

MS_PERSONAL = RegisterWindowMessage("MS_PERSONAL")

Aqui registro un nuevo mensaje, y almaceno dicho ID en la variable MS_PERSONAL, luego uso dicho ID en el gestor de mensaje, al igual como hemos tratado todos los mensajes. Creo que es sencillo ver lo que hace.

ABRIENDO VENTANAS

Para este proyecto necesitamos un modulo .bas, y dos formularios los cuales tendrán estos nombres: frmDraw y frmDraw1, en el primero colocaran un botón y en las propiedades del formulario “StartUpPosition” coloquen “CenterScreen”, luego copian este código:

En el modulo:

Declare Function SetRect Lib "User32" (lpRect As RECT, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Declare Function DrawAnimatedRects Lib "User32" (ByVal hWnd As Long, ByVal idAni As Long, lprcFrom As RECT, lprcTo As RECT) As Long

Public Const IDANI_CAPTION = &H3

Private Type RECT

    Left As Long

    Top As Long

    Right As Long

    Bottom As Long

End Type

Public recf1s As RECT

Public recf1d As RECT

En frmDraw:

Private Sub Command1_Click()

    frmDraw1.Show

End Sub

En frmDraw1:

Private Sub Form_Load()

swidth = Screen.Width / Screen.TwipsPerPixelX

sheight = Screen.Height / Screen.TwipsPerPixelY

SetRect recf1s, swidth, sheight, swidth, sheight

SetRect recf1d, 100, 300, 200, 200

DrawAnimatedRects Me.hWnd, IDANI_CAPTION, recf1s, recf1d

Me.Move 100 * Screen.TwipsPerPixelX, 300 * Screen.TwipsPerPixelY, 100 * Screen.TwipsPerPixelX, 100 * Screen.TwipsPerPixelY

End Sub

Luego arrancan el programa y presionan el botón, ¿Ven el efecto?. No lo ven!!,  cierren el formulario y dale al botón de nuevo,  todavía no ves el efecto, colócale comentario a la línea:

DrawAnimatedRects Me.hWnd, IDANI_CAPTION, recf1s, recf1d

Arranca y dale al botón,  veras que antes el formulario hacia un efecto animado de abajo hacia arriaba, y ahora tan sencillamente abre el formulario sin tal efecto, si no lo ves todavía quita el comentario y ejecútalo, ¡fui fastidioso! ya que yo tuve que arrancar varias veces el programa para ver el efecto.

Las API’S a utilizar son sencilla y las describo a continuación:

SetRect

Api: Declare Function SetRect Lib "User32" (lpRect As RECT, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Esta API es simple lo único que hace es llenar una variable de tipo RECT. Donde X1 es el equivalente a Left, Y1 a Top, X2  a Right, y Y2 a Bottom.

DrawAnimatedRects

Api: Declare Function DrawAnimatedRects Lib "User32" (ByVal hWnd As Long, ByVal idAni As Long, lprcFrom As RECT, lprcTo As RECT) As Long

Esta función es utilizada para realizar el efecto de desplazamiento desde un lugar de origen hasta un lugar destino.

Como primer parámetro le pasamos el “hwdn” de la ventana a la cual queremos aplicar dicho efecto. El segundo parámetro especifica el tipo de animación, en la documentación oficial, solo veo una sola constante que es IDANI_CAPTION , como tercer parámetro le pasamos una variable de tipo RECT que determina el rectángulo de donde partirá la ventana, y como último parámetro pasamos otra variable de tipo RECT que determina el rectángulo destino.

En nuestro código tenemos:

SetRect recf1s, swidth, sheight, swidth, sheight ‘ Esta es la esquina inferior derecha de la pantalla

SetRect recf1d, 100, 300, 200, 200

DrawAnimatedRects Me.hWnd, IDANI_CAPTION, recf1s, recf1d

Es decir, animamos desde la esquina inferior derecha de la pantalla, hasta el punto 100,300, y la damos al formulario una dimensión de 200x200.

OTRAS FUNCIONES

CloseWindow

Api: Declare Function CloseWindow Lib "user32" Alias "CloseWindow" (ByVal hwnd As Long) As Long

Esta función se encarga de minimizar la ventana, especificada en el parámetro “hwnd”, OJO! Minimiza, mas no destruye.

EnableWindow

Api: Declare Function EnableWindow Lib "user32" Alias "EnableWindow" (ByVal hwnd As Long, ByVal fEnable As Long) As Long

Función encargada de habilitar o deshabilitar a la ventana, para que reciba los mensajes de mouse y teclado. Como primer parámetro se le pasa “hwnd” de la ventana, y como segundo parámetro se le pasa TRUE o FALSE, en caso de pasarle TRUE implica que la ventana esta activada, por lo tanto recibe los entradas de mouse y teclado, en caso contrario se le pasaría FALSE para no permitir dichas entradas.

FlashWindowEx

Api: Declare Function FlashWindowEx Lib "user32" (pfwi As FLASHWINFO) As Boolean

Esta función hace que el caption de la ventana se ponga intermitente, es parecido cuando estamos en una ventana, y otra aplicación requiere de atención del usuario, esta empieza a titilar, avisando que necesita atención, esa efecto de titilar se logra con esta función.

Como parámetro tiene una estructura de esta forma:

Private Type FLASHWINFO

    cbSize As Long

    hwnd As Long

    dwFlags As Long

    uCount As Long

    dwTimeout As Long

End Type
CbSize = es el tamaño de la estructura, para ello apliquen la función Len() de VB.

Hwnd = es el hwnd de la ventana a la cual queremos aplicar el Flash.

DwFlags = especifica el status de dicho flash, puede tomar los siguientes valores:

Public Const FLASHW_STOP = 0 : Si el flash se encuentra activado, al usar este parámetro se desactiva.

Public Const FLASHW_CAPTION = &H1 : Con este parámetro el caption de la ventana empieza a titilar (flash).

Public Const FLASHW_TRAY = &H2 : Con este parámetro titila el botón ubicado en el “TaskBar Button”, es decir, el botón de la ventana que esta en la barra de herramientas donde esta el botón “inicio”

Public Const FLASHW_ALL = (FLASHW_CAPTION Or FLASHW_TRAY) : Es la combinacion de los 2 ya mencionados.

Public Const FLASHW_TIMER = &H4 : El flash o titilar, se mantinen continuo hasta que se especifique FLASHW_STOP

Public Const FLASHW_TIMERNOFG = &HC : El flash es continuo hasta seleccionar la ventana.

UCount = especifica el número de veces que titilara la ventana.

DwTimeout = especifica el rango, en milisegundos, en el cual la ventana empezara a titilar. Puede ser 0, en este caso el sistema usa un rango por defecto.

IsChild

Api: Declare Function IsChild Lib "user32" Alias "IsChild" (ByVal hWndParent As Long, ByVal hwnd As Long) As Long

Función que determina si una ventana es hija de una ventana padre, como primer parámetro pasamos el “hwnd” de la ventana padre a evaluar, y como segundo parámetro pasamos la supuesta ventana hija que queremos comprobar.

Si la ventana hija se encuentra en la ventana padre, la función retorna un valor distinto de 0, en caso de que la ventana no sea hija retorna el valor 0.

IsIconic

Api: Declare Function IsIconic Lib "user32" Alias "IsIconic" (ByVal hwnd As Long) As Long

Función que determina si la ventana esta minimizada, como único parámetro toma el “hwnd” que queremos evaluar. Si la función retorna distinto de 0 implica que la ventana esta minimizada, en caso contrario retorna 0.

IsWindowEnabled

Api: Declare Function IsWindowEnabled Lib "user32" Alias "IsWindowEnabled" (ByVal hwnd As Long) As Long

Función que determina si la ventana especificada en el primer parámetro, se encuentra activada “enabled” o desactivada “disabled”, a las entradas del mouse y teclado.

Si la ventana esta activada, el valor que retorna es 0, en caso contrario retornara un valor distinto de 0.

IsZoomed

Api: Declare Function IsZoomed Lib "user32" Alias "IsZoomed" (ByVal hwnd As Long) As Long

Determina si la ventana pasada en el primer parámetro, esta maximizada. En caso de ser así, la función retornara un valor distinto de 0, en caso contrario retornara 0.

LockWindowUpdate

Api: Declare Function LockWindowUpdate Lib "user32" Alias "LockWindowUpdate" (ByVal hwndLock As Long) As Long

Función que bloquea el pintado de la ventana, es decir, el mensaje WM_PAINT es bloqueado. Para bloquear dicha ventana, se le pasa su “hwnd” en el primer parámetro, para volver a activar dicha ventana se le pasa FALSE en el primer parámetro.

MoveWindow

Api: Declare Function MoveWindow Lib "user32" Alias "MoveWindow" (ByVal hwnd As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal bRepaint As Long) As Long

Función que se encarga de mover la ventana pasada en el primer parámetro. Del segundo al cuarto parámetro se establece la nueva posición y el nuevo tamaño. Como sexto parámetro “bRepaint”, la pasamos TRUE si queremos que la ventana se repinta una vez movida. O FALSE en caso contrario.

En conclusión ¿Para que sirve todo lo visto en este Capítulo ?, específicamente ¿Para que crear una ventana de la forma indicada en C y en el primer proyecto de dicho Capítulo ?

Ciertamente, puede que muchos programadores no usen nada de esto en el transcurso de su carrera, pero uno nunca que sabe que problema nos depara el destino. Vamos a poner un ejemplo que me paso a mi.

Un día estaba desarrollando una aplicación en VB; después que finalizo el proyecto, le hice al cliente un instalador, ese instalador lo hice usando un software que genera dicho instaladores, por lo que la instalación no queda de parte mía, sino del software que se encarga de ello. Cuando el cliente ve la instalación, le pareció perfecto, “PERO”, siempre el usuario tiene que salir con un “pero”, el cliente quería que cuando el usuario introdujera el CD-ROM el programa detectara si estaba instalado el programa, en caso de estarlo, arrancar dicho programa, en caso contrario ejecutar la instalación.

Ahí teníamos un serio problema ¿Por qué?. Porque ¿Cómo hacia el programa? Ustedes dirán ¡PERO SI ES FACIL!, usas VB, haces un formulario que detecte el programa y LISTO!!. Y yo les digo: ¿Y como arrancan ese programa si no tienen las DLL necesarias para arrancar la aplicación?. Todos sabemos que para arrancar una aplicación mínima en VB se necesita que la maquina del usuario posea ciertos archivos DLL, pero entonces algunos dirán:  Sigue siendo fácil, haces un programa que copie las DLL. ¿Y con que lenguaje hacen ese programa?. Si lo hacen en VB quedan en el mismo problema.

Para solucionar eso, tuve que hacer un pequeño programa en C similar al escrito en el Capítulo  II,  el programa se encargaba de detectar si la aplicación se encontraba instalada, era relativamente un programa pequeño, por lo que no me complique mucho con las API. ¿Y ese programa funciona en todas las versiones de Windows, no necesita DLL?. Por supuesto que necesita DLL, pero son DLL que ya vienen con Windows, y funcionara en todas las versiones, siempre y cuando usen funciones API’S compatibles con todas las versiones. Con ese programa solucione el problema, sabiendo que siempre iba a funcionar, sin necesidad de instalar ningún archivo extra.

Capítulo  VI.  Menús
Los menús deben de ser algo ya conocidos por ustedes, pero en este Capítulo  vamos a tratar funciones API que se encargan de la creación de un menú desde 0, es decir, si usar la utilidad que nos presenta VB para la creación de los mismos, ¿Y para que aprender eso?. Ciertamente es algo que muy pocas veces y tal vez nunca utilicen pero siempre es bueno saber como funcionan los componentes o recursos de Windows, por que estos nos ayuda a obtener mayor control sobre nuestro software, y veremos un proyecto muy interesante, donde la utilización de las funciones son primordiales ya que el editor de menú de VB no permite hacer lo que se muestra en el proyecto.

¿Qué es menú?. Se puede decir, que un menú es una lista de elementos o comandos, los cuales son desplegados de una manera ordenada, dando al usuario la opción de elegir uno de esos elementos, solamente pudiendo elegir un elemento a la vez.

Pero ese menú esta formado por una barra de menú “menú bar”, que es el menú principal, este es el que se encarga de desplegar otros menús que se llaman “sub-menús”. Todos estos elementos que conforman al menú se encuentran colocados en un orden de posición como se muestra a continuación:









Con estos números les indico la posición donde se encuentra cada elemento “item”, del menú. Llamamos barra de menú, aquella que contiene los elementos “Archivo”, “Edición” y “Salir”, y llamamos sub-menús, al que contiene “Nuevo”, “una línea de separación”, “Guardar”, “Cerrar”.

Ustedes dirían ¿Por qué hay dos 0, por que hay dos 1, por que no continuar con la secuencia de numeración?. Resulta que aquí ustedes creen que ven UN solo menú, pero a nivel de programación hay 2 menús, el primer menú es la barra de menús, y el segundo es un submenú. Es decir, hay 2 objetos menús en este ejemplo básico. Lo que sucede es que pensamos que todo es un menú, ciertamente es así, pero a nivel de programación se trabaja por separado. Ya lo veremos mejor en el primer proyecto.

Ya mencione que un menú, posee elementos “item”, los cuales generan una acción “command”, esa acción deberá de ser procesada por nuestro programa para hacer algo, por ejemplo, “Nuevo” es un elemento que puede realizar la acción de crear un documento, o imagen, etc. ¿Desde ese punto de vista “Archivo”, es un elemento que genera una acción?. Cuando un elemento lleva asociado un submenú, este elemento deja de generar una acción “command”, ya que la acción la procesa directamente Windows, y dicha acción sería desplegar el submenú asociado a dicho elemento.

Volviendo a los item, cada item, es como un pequeño objeto que puede tener diferentes estados, como desactivado, seleccionado, apagado, marcado, etc. Cuando vayamos a crear dicho elemento vamos a necesitar una estructura del tipo MENUITEMINFO, esta estructura contiene toda la configuración posible que se le puede asignar a dicho elemento, cada elemento que queramos añadir será una variable del tipo MENUITEMINFO, totalmente independiente.

Por esa razón procedo a explicar dicha estructura y sus diferentes constantes antes de entrar al primer proyecto.

La estructura esta configurada de la siguiente manera:

Public Type MENUITEMINFO

    cbSize As Long

    fMask As Long

    fType As Long

    fState As Long

    wID As Long

    hSubMenu As Long

    hbmpChecked As Long

    hbmpUnchecked As Long

    dwItemData As Long

    dwTypeData As String

    cch As Long

End Type
CbSize = es el tamaño de dicha estructura se valor se obtiene con la función Len() de VB.

FMask = puede tomar las constantes que se presentan a continuación:

Public Const MIIM_CHECKMARKS = &H8 : Permite asignar o recuperar los valores correspondiente a las variables hbmpChecked y hbmpUnChecked.
Public Const MIIM_DATA = &H20 : Permite asignar o recuperar el valor correspondiente de la variable dwItemData.
Public Const MIIM_ID = &H2 : Permite asignar o recuperar el valor correspondiente de la variable wID.
Public Const MIIM_STATE = &H1 : Permite asignar o recuperar el valor correspondiente de la variable fState
Public Const MIIM_SUBMENU = &H4 : Permite asignar o recuperar el valor correspondiente de la variable hSubMenu
Public Const MIIM_TYPE = &H10 : Permite asignar o recuperar el valor correspondiente de la variable dwTypeData y fType.
El como se usa esta parámetro fMask lo veremos cuando explique el código del primer proyecto.

Ftype = este parámetro se utiliza para configurar al elemento del menú, y puede tomar las siguientes constantes:

Public Const MFT_BITMAP = &H4& : Con esto indicamos que el elemento del menú, será bitmap. El bitmap debe de ser pasado en la parte baja “low-order” de la variable dwTypeData. Y el parámetro “cch” es ignorado.
Public Const MFT_MENUBARBREAK = &H20& : Este parámetro coloca al elemento en una nueva línea en caso de la barra de menú, o en una nueva columna en caso de ser un sub-menú, y dicha columna es separada por una línea vertical, es decir, un submenú no tiene por que ser un listado completamente vertical, pueden haber varias columnas con elementos. En el proyecto lo verán un poco mejor.

Public Const MFT_MENUBREAK = &H40& : Hace lo mismo que la constante antes explicada, con la diferencia que la línea vertical que separa a dicha columna no es dibujada.

Public Const MFT_OWNERDRAW = &H100& : Con esta constante especificamos que vamos hacer nosotros lo que vamos a pintar el menú. Para esta constante existe un proyecto que veremos mas adelante.

Public Const MFT_RADIOCHECK = &H200& : Despliega un circulo negro al lado del elemento en caso de estar seleccionado, si el parámetro de la estructura hbmpChecked, tiene una imagen bitmap, pues se mostrara en ves del circulo dicho bitmap.

Public Const MFT_RIGHTJUSTIFY = &H4000& : Justifica el texto hacia la derecha, solo puede ser usado en la barra de menú “menú bar”

Public Const MFT_RIGHTORDER = &H2000& : 

Public Const MFT_SEPARATOR = &H800& : Indica que el elemento es una línea de separación.

Public Const MFT_STRING = &H0& : Indica que el elemento viene representado por una cadena de caracteres.

Para este parámetro “fType” pueden usar varias constantes, pero con lógica, es decir, es ilógico usar “MFT_BITMAP or MFT_STRING”, ya que el elemento o es un bitmap o es una cadena de caracteres, pero no puede ser las dos cosas juntas.

FState: con esto indicamos o modificamos el estado de una elemento, puede tomar las siguientes constantes: 

Public Const MFS_CHECKED = &H8& : configura al elemento como “seleccionado” o activo, es decir, coloca el circulo negro al lado del texto, o coloca la imagen pasada en el parámetro hbmpChecked
Public Const MFS_DEFAULT = &H1000& : Especifica que dicho elemento, es el elemento por defecto, y lo marca con letras negritas.

Public Const MFS_GRAYED = &H3& : Desactiva el elemento del menú, colocando el texto en gris claro.

Public Const MFS_DISABLED = MFS_GRAYED : Hace lo mismo que MFS_GRAYED.

Public Const MFS_ENABLED = &H0& : Activa el elemento del menú, regresándolo a su estado original.

Public Const MFS_HILITE = &H80& : Marca en azul al elemento cuando este visible. ¿Qué es eso de azul?. Bueno el azul dependerá de la configuración de los colores de Windows. Cuando ustedes despliegan un menú y pasan el mouse por encima de los elementos, este se marca, en mi maquina y en la configuración por defecto se marca en azul. A eso es lo que me refiero.

Public Const MFS_UNCHECKED = &H0& : configura al elemento como “deseleccionado” o desactivado, es decir, quita el circulo negro al lado del texto, o coloca la imagen pasada en el parámetro hbmpUnChecked
Public Const MFS_UNHILITE = &H0& : Si MFS_HILITE marcaba como seleccionado al elemento, es decir, lo pintaba de “azul” pues este parámetro hace lo contrario.

WID = Es un número de 16 bit único que identifica a dicho elemento.

HSubMenu = si dicho elemento al ser seleccionado despliega un submenú, pues aquí se pasa el “handle” de dicho menú. En caso de ser un elemento que realiza una acción “command”, pues se le coloca 0.

hbmpChecked = aquí le pasamos la imagen a mostrar en caso de que se haya definido en el parámetro fType la constante MFT_RADIOCHECK. Esta imagen es colocada al lado izquierdo del texto mostrado. Y dicha imagen es mostrada cuando el status del elemento esta como “seleccionado”
hbmpUnchecked =  aquí le pasamos la imagen a mostrar en caso de que se haya definido en el parámetro fType la constante MFT_RADIOCHECK. Esta imagen es colocada al lado izquierdo del texto mostrado. Y dicha imagen es mostrada cuando el status del elemento esta como “deseleccionado”
dwItemData = Valor asociado al elemento del menú, en realidad, particularmente yo no le he visto uso a este parámetro.

DwTypeData = Se puede asignar información en dicho parámetro, cuando en fMask se especifica la constante MIIM_TYPE, este parámetro en caso de que en “ftype” se coloque la constante MFT_STRING, va la cadena de caracteres a mostrar en el menú. En caso de que sea MFT_BITMAP la parte baja de esta variable “low-order word”, tendrá la imagen a mostrar.

Cch = En caso de que se haya configurado MFT_STRING, en este parámetro va la longitud de la cadena pasada en “dwTypeData”.

Esta estructura como pueden ver es muy completa, y refleja la configuración de los elementos de un menú. Esta estructura es utilizada en todas las funciones que involucren la manipulación de los elementos de un menú, en caso de haber quedado confundidos en el uso de algunos parámetros no se preocupen que en el transcurso del Capítulo  aclararemos algunos de estos parámetro.

CREANDO MENUS

Ya explicado la configuración de los elementos de un menú, vamos a proceder a crear un menú, usando puras funciones API’S es decir, no vamos a usar el editor de menús, que presenta VB, con esto entenderemos un poco mejor el funcionamiento interno de los menús en un programa.

Para este proyecto necesitan un modulo y un formulario, en el formulario van a colocar 2 controles Frame, y un Label, dentro del frame1, colocan 3 botones, y en el otro frame2 colocan 3 botones mas. Déjenles los nombres por defecto a los controles. Pero al formulario lo van a llamar “frmMenu”.

Tambien busquen en su computadora si poseen este archivo “TLBINF32.dll” en caso de no tenerlo, en el zip que contiene dicho documento, se encuentra el archivo, el cual debe de ser copiado en la carpeta Window-System.

En el modulo colocan este código:

Option Explicit

Public Const MIIM_CHECKMARKS = &H8

Public Const MIIM_DATA = &H20

Public Const MIIM_ID = &H2

Public Const MIIM_STATE = &H1

Public Const MIIM_SUBMENU = &H4

Public Const MIIM_TYPE = &H10

Public Const MFT_BITMAP = &H4&

Public Const MFT_MENUBARBREAK = &H20&

Public Const MFT_MENUBREAK = &H40&

Public Const MFT_RADIOCHECK = &H200&

Public Const MFT_RIGHTJUSTIFY = &H4000&

Public Const MFT_SEPARATOR = &H800&

Public Const MFT_STRING = &H0&

Public Const MFS_CHECKED = &H8&

Public Const MFS_DEFAULT = &H1000&

Public Const MFS_ENABLED = &H0&

Public Const MFS_UNCHECKED = &H0&

Public Const IMAGE_ICON = 1

Public Const IMAGE_BITMAP = 0

Public Const LR_LOADFROMFILE = &H10

Public Type RECT

    Left As Long

    Top As Long

    Right As Long

    Bottom As Long

End Type

Public Type MENUITEMINFO

    cbSize As Long

    fMask As Long

    fType As Long

    fState As Long

    wID As Long

    hSubMenu As Long

    hbmpChecked As Long

    hbmpUnchecked As Long

    dwItemData As Long

    dwTypeData As String

    cch As Long

End Type

Declare Function CreateMenu Lib "user32" () As Long

Declare Function InsertMenuItem Lib "user32" Alias "InsertMenuItemA" (ByVal hMenu As Long, ByVal un As Long, ByVal bool As Boolean, lpcMenuItemInfo As MENUITEMINFO) As Long

Declare Function DeleteMenu Lib "user32" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long) As Long

Declare Function DestroyMenu Lib "user32" (ByVal hMenu As Long) As Long

Declare Function DrawMenuBar Lib "user32" (ByVal hwnd As Long) As Long

Declare Function SetMenu Lib "user32" (ByVal hwnd As Long, ByVal hMenu As Long) As Long

Declare Function GetMenuItemInfo Lib "user32" Alias "GetMenuItemInfoA" (ByVal hMenu As Long, ByVal un As Long, ByVal b As Boolean, lpMenuItemInfo As MENUITEMINFO) As Long

Declare Function SetMenuItemInfo Lib "user32" Alias "SetMenuItemInfoA" (ByVal hMenu As Long, ByVal un As Long, ByVal bool As Boolean, lpcMenuItemInfo As MENUITEMINFO) As Long

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal un1 As Long, ByVal n1 As Long, ByVal n2 As Long, ByVal un2 As Long) As Long

Declare Function loword Lib "TLBINF32" (ByVal DWord As Long) As Integer

Declare Function hiword Lib "TLBINF32" (ByVal DWord As Long) As Integer

Public Const GWL_WNDPROC = (-4)

Const WM_COMMAND = &H111

Const WM_MENUSELECT = &H11F

Dim ViejoGestor As Long

Public entro As Boolean

Public Const ID_ARCHIVO = 1

Public Const ID_EDICION = 2

Public Const ID_SALIR = 3

Public Const ID_HERRAMIENTA = 4

Public Const ID_NUEVO = 5

Public Const ID_BARRA = 6

Public Const ID_SELECCION = 7

Public Const ID_SELECCION2 = 8

Public Const ID_MENU1 = 1

Public Const ID_MENU2 = 2

Public hMenu1 As Long

Public hMenu2 As Long

Public Archivo As MENUITEMINFO

Public Edicion As MENUITEMINFO

Public Herramienta As MENUITEMINFO

Public Salir As MENUITEMINFO

Public Nuevo As MENUITEMINFO

Public Barra As MENUITEMINFO

Public Seleccion As MENUITEMINFO

Public Seleccion2 As MENUITEMINFO

Public inf As MENUITEMINFO

Public Sub nuevoGestor(hwnd As Long)

    ViejoGestor = SetWindowLong(hwnd, GWL_WNDPROC, AddressOf GestorMensaje)

End Sub

Public Function GestorMensaje(ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    If mensaje = WM_MENUSELECT Then

        Select Case loword(wParam)

        Case ID_NUEVO

            frmMenu.Label1 = "ESTAS EN NUEVO"

        Case ID_SELECCION

            frmMenu.Label1 = "ESTAS EN SELECCCION"

        Case ID_SELECCION2

            frmMenu.Label1 = "ESTAS EN SELECCCION 2"

        End Select

    End If

    If mensaje = WM_COMMAND Then

        Select Case wParam

        Case ID_SALIR

            frmMenu.SalirAplicacion

        Case ID_SELECCION

            inf.cbSize = Len(inf)

            inf.fMask = MIIM_STATE

            GetMenuItemInfo hMenu2, ID_SELECCION, False, inf

            If inf.fState = MFS_CHECKED Then

                inf.fState = MFS_UNCHECKED

            Else

                inf.fState = MFS_CHECKED

            End If

            SetMenuItemInfo hMenu2, ID_SELECCION, False, inf

        Case ID_SELECCION2

            inf.cbSize = Len(inf)

            inf.fMask = MIIM_STATE

            GetMenuItemInfo hMenu2, ID_SELECCION2, False, inf

            If inf.fState = MFS_CHECKED Then

                inf.fState = MFS_UNCHECKED

            Else

                inf.fState = MFS_CHECKED

            End If

            SetMenuItemInfo hMenu2, ID_SELECCION2, False, inf

        End Select

    End If

    GestorMensaje = CallWindowProc(ViejoGestor, hwnd, mensaje, wParam, lParam)

End Function

En el formulario este código:

Option Explicit

Private Sub Command1_Click()

    CreacionBarraMenu

End Sub

Sub CreacionBarraMenu()

    hMenu1 = CreateMenu()

    InicializarEstructurasMenuBar

    InsertMenuItem hMenu1, 0, True, Archivo

    InsertMenuItem hMenu1, 1, True, Edicion

    InsertMenuItem hMenu1, 2, True, Salir

    SetMenu Me.hwnd, hMenu1

End Sub

Sub InicializarEstructurasMenuBar()

    Archivo.cbSize = Len(Archivo)

    Archivo.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Archivo.fType = MFT_STRING

    Archivo.fState = MFS_ENABLED

    Archivo.wID = ID_ARCHIVO

    Archivo.hSubMenu = 0

    Archivo.dwItemData = 0

    Archivo.dwTypeData = "Archivo"

    Archivo.cch = Len("Archivo")

    Archivo.hbmpChecked = 0

    Archivo.hbmpUnchecked = 0

    Edicion.cbSize = Len(Edicion)

    Edicion.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Edicion.fType = MFT_STRING

    Edicion.fState = MFS_ENABLED

    Edicion.wID = ID_EDICION

    Edicion.hSubMenu = 0

    Edicion.dwItemData = 0

    Edicion.dwTypeData = "Edicion"

    Edicion.cch = Len("Edicion")

    Edicion.hbmpChecked = 0

    Edicion.hbmpUnchecked = 0

    Herramienta.cbSize = Len(Herramienta)

    Herramienta.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Herramienta.fType = MFT_STRING

    Herramienta.fState = MFS_ENABLED

    Herramienta.wID = ID_HERRAMIENTA

    Herramienta.hSubMenu = 0

    Herramienta.dwItemData = 0

    Herramienta.dwTypeData = "Herramienta"

    Herramienta.cch = Len("Herramienta")

    Herramienta.hbmpChecked = 0

    Herramienta.hbmpUnchecked = 0

    Salir.cbSize = Len(Salir)

    Salir.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Salir.fType = MFT_STRING

    Salir.fState = MFS_DEFAULT

    Salir.wID = ID_SALIR

    Salir.hSubMenu = 0

    Salir.dwItemData = 0

    Salir.dwTypeData = "Salir"

    Salir.cch = Len("Salir")

    Salir.hbmpChecked = 0

    Salir.hbmpUnchecked = 0

End Sub

Private Sub Command2_Click()

    SetMenu Me.hwnd, 0

    DestroyMenu hMenu1

End Sub

Private Sub Command3_Click()

    InsertMenuItem hMenu1, 1, True, Herramienta

    DrawMenuBar Me.hwnd

End Sub

Private Sub Command4_Click()

    SetMenu Me.hwnd, 0

    DestroyMenu hMenu1

End Sub

Private Sub Command5_Click()

    CreaMenuSubMenu

End Sub

Sub CreaMenuSubMenu()

    hMenu1 = CreateMenu()

    hMenu2 = CreateMenu()

    InicializarEstructurasMenuBar

    InicializarEstructurasSubMenu

    InsertMenuItem hMenu2, 0, True, Nuevo

    InsertMenuItem hMenu2, 1, True, Barra

    InsertMenuItem hMenu2, 2, True, Seleccion

    InsertMenuItem hMenu2, 3, True, Seleccion2

    Archivo.fMask = Archivo.fMask Or MIIM_SUBMENU

    Archivo.hSubMenu = hMenu2

    InsertMenuItem hMenu1, 0, True, Archivo

    InsertMenuItem hMenu1, 1, True, Edicion

    InsertMenuItem hMenu1, 2, True, Salir

    SetMenu Me.hwnd, hMenu1

End Sub

Sub InicializarEstructurasSubMenu()

    Nuevo.cbSize = Len(Nuevo)

    Nuevo.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Nuevo.fType = MFT_STRING

    Nuevo.fState = MFS_ENABLED

    Nuevo.wID = ID_NUEVO

    Nuevo.hSubMenu = 0

    Nuevo.dwItemData = 0

    Nuevo.dwTypeData = "&Nuevo"

    Nuevo.cch = Len("Nuevo")

    Nuevo.hbmpChecked = 0

    Nuevo.hbmpUnchecked = 0

    Barra.cbSize = Len(Barra)

    Barra.fMask = MIIM_TYPE

    Barra.fType = MFT_SEPARATOR

    Barra.fState = 0

    Barra.wID = ID_BARRA

    Barra.hSubMenu = 0

    Barra.dwItemData = 0

    Barra.dwTypeData = 0

    Barra.cch = 0

    Barra.hbmpChecked = 0

    Barra.hbmpUnchecked = 0

    Seleccion.cbSize = Len(Seleccion)

    Seleccion.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    'Seleccion.fType = MFT_MENUBARBREAK

    'Seleccion.fType = MFT_MENUBREAK

    Seleccion.fType = MFT_RADIOCHECK

    Seleccion.fState = MFS_ENABLED Or MFS_CHECKED

    Seleccion.wID = ID_SELECCION

    Seleccion.hSubMenu = 0

    Seleccion.dwItemData = 0

    Seleccion.dwTypeData = "Seleccion"

    Seleccion.cch = Len("Seleccion")

    Seleccion.hbmpChecked = 0

    Seleccion.hbmpUnchecked = 0

    Seleccion2.cbSize = Len(Seleccion)

    Seleccion2.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE Or MIIM_CHECKMARKS

    Seleccion2.fType = MFT_RADIOCHECK

    Seleccion2.fState = MFS_ENABLED Or MFS_CHECKED

    Seleccion2.wID = ID_SELECCION2

    Seleccion2.hSubMenu = 0

    Seleccion2.dwItemData = 0

    Seleccion2.dwTypeData = "Seleccion 2"

    Seleccion2.cch = Len("Seleccion 2")

    Seleccion2.hbmpChecked = LoadImage(App.hInstance, App.Path & "\factura.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

    Seleccion2.hbmpUnchecked = LoadImage(App.hInstance, App.Path & "\printer.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

End Sub

Private Sub Form_Load()

    nuevoGestor Me.hwnd

    Command1.Caption = "Crear Menu"

    Command1.Height = 510

    Command1.Left = 180

    Command1.Top = 360

    Command1.Width = 870

    Command2.Caption = "Quitar Menu"

    Command2.Height = 510

    Command2.Left = 2070

    Command2.Top = 360

    Command2.Width = 870

    Command3.Caption = "Añadir Menu"

    Command3.Height = 510

    Command3.Left = 1125

    Command3.Top = 360

    Command3.Width = 870

    Command4.Caption = "Quitar Menu"

    Command4.Height = 510

    Command4.Left = 1125

    Command4.Top = 360

    Command4.Width = 870

    Command5.Caption = "Crear Menu"

    Command5.Height = 510

    Command5.Left = 180

    Command5.Top = 360

    Command5.Width = 870

    Frame1.Caption = "Barra de menú Simple"

    Frame1.Height = 1050

    Frame1.Top = 90

    Frame1.Left = 45

    Frame1.Width = 3075

    Frame2.Caption = "Barra de menú Compleja"

    Frame2.Height = 1050

    Frame2.Top = 90

    Frame2.Left = 3150

    Frame2.Width = 2130

    Label1.Caption = ""

    Label1.Height = 510

    Label1.Left = 45

    Label1.Top = 1170

    Label1.Width = 2445

    Me.Height = 2280

    Me.Width = 5670

End Sub

Private Sub Form_Paint()

    SetMenu Me.hwnd, hMenu1

    DrawMenuBar Me.hwnd

End Sub

Private Sub Form_Unload(Cancel As Integer)

    DestroyMenu hMenu1

    DestroyMenu hMenu2

End Sub

Sub SalirAplicacion()

    Unload Me

End Sub

Es claro ver que estamos usando la técnica estudiada en el Capítulo  anterior llamada SubClassing, en este caso se lo aplicamos al menú. ¿Por qué usamos SubClassing?. Como estamos creando un menú sin usar el editor de VB, los mensajes hay que capturarlos “manualmente”, es decir, tenemos nosotros que interceptar el mensaje en el gestor de mensajes del formulario.

Se corren el programa verán, que un frame corresponde a un menú muy básico, es decir, solamente se crea la barra de menú, y el otro frame crea un menú un poco mas complejo ya que el elemento “Archivo”, posee internamente un Submenú. Vamos a ver ahora las funciones encargadas de dicha creación.

CreateMenu

Api: Declare Function CreateMenu Lib "user32" () As Long

Esta función es sencilla y lo que hace es crear un objeto del tipo menú. Si la creación es un éxito el valor que retorna es distinto de 0.

En nuestro proyecto usamos esta función, en el frame de “barra de menú simple”; el botón de “crear menú” llama a una función llamada “CreacionBarraMenu” aquí tenemos esta línea:

hMenu1 = CreateMenu()

Es decir, hMenu1 es nuestra variable del tipo “menú”. En el otro frame, en el botón “crear menu”  tenemos 2 líneas:

hMenu1 = CreateMenu()

hMenu2 = CreateMenu()

¿Por qué 2 menús? Recuerden que la barra de menú es un “menú”, y el Submenú que abre cuando se presiona en “Archivo” es otro “menú”. Ambos independientes uno del otro.

DestroyMenu

Api: Declare Function DestroyMenu Lib "user32" (ByVal hMenu As Long) As Long

Todo lo que se crea a “mano” debe de ser destruido y esta función es la que se encarga de destruir los objetos del tipo menú. Como primer parámetro se le pasa el “handle” del menú que se obtuvo en la creación del mismo, en pocas palabras se pasan en el primer parámetro las variables utilizados para la creación de los menús, que en nuestro caso son hMenu1 y hMenu2.

NOTA

Antes de continuar, vamos a explicar un poco lo que hace la función InicializarEstructurasMenuBar  en esta función es donde se configuran los elementos que conformaran en este caso a la barra de menú, y la función InicializarEstructurasSubMenu es la encargada de configurar los elementos que conforman al submenú que se habilita en “Archivo”, voy a explicar un poco como configurar los elementos y la explicación de uno vale para todos los demás.

Vamos a ver como se configura el elemento “Archivo”.

    Archivo.cbSize = Len(Archivo)

    Archivo.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Archivo.fType = MFT_STRING

    Archivo.fState = MFS_ENABLED

    Archivo.wID = ID_ARCHIVO

    Archivo.hSubMenu = 0

    Archivo.dwItemData = 0

    Archivo.dwTypeData = "Archivo"

    Archivo.cch = Len("Archivo")

    Archivo.hbmpChecked = 0

    Archivo.hbmpUnchecked = 0

El parámetro de la estructura que mas puede confundir es “fMask”, los valores de este parámetro dependerá de lo que vayamos a configurar, es decir, en esta estructura configuramos los parámetros “fType”, “fState” y “wID”, para que sea valido la asignación de dichos valores, es necesario indicarle que estamos cambiando esas variables y lo hacemos aplicando un “or” lógico a las constantes MIIM_ID Or MIIM_STATE Or MIIM_TYPE, por ejemplo, si quitáramos la constante 

MIIM_TYPE el menú no muestra nada ¿Por que no muestra nada si colocamos la cadena en “dwTypeData”?. Recuerden la definición de la constante MIIM_TYPE: “Permite asignar o recuperar el valor correspondiente de la variable dwTypeData y fType”. Cuando nosotros llenamos el valor de “dwTypeData”, si queremos que ese valor se le asigne realmente al elemento es necesario especificar en “fMask” la constante que permite que las funciones que manipulan a los elementos tomen en cuenta dicho valor.

Vean por ejemplo como se configura la barra separable:

    Barra.cbSize = Len(Barra)

    Barra.fMask = MIIM_TYPE

    Barra.fType = MFT_SEPARATOR

    Barra.fState = 0

    Barra.wID = ID_BARRA

    Barra.hSubMenu = 0

    Barra.dwItemData = 0

    Barra.dwTypeData = 0

    Barra.cch = 0

    Barra.hbmpChecked = 0

    Barra.hbmpUnchecked = 0

Lo único que configuramos aquí es el “fType”, ya que todos los demás valores no importa, cuando ponemos en “fMask”, MIIM_TYPE, en este caso toma en cuenta “fType” y omite todos los demás parámetros. Ustedes podrían decir ¿Para que entonces configuras “wID”?. Lo hago para seguir un patrón,  el parámetro “wID” sirve al momento de capturar los mensajes, pero como en una barra de separación no nos interesa capturar el mensaje ya que ella no es un elemento de acción. Pues no importa el valor que pongamos, en este caso es omitido.

InsertMenuItem

Api: Declare Function InsertMenuItem Lib "user32" Alias "InsertMenuItemA" (ByVal hMenu As Long, ByVal un As Long, ByVal bool As Boolean, lpcMenuItemInfo As MENUITEMINFO) As Long

Ya creado un menú con CreateMenu, a ese menú le tenemos que insertar “elementos” y esta función es la que se encarga de ello.

Como primer parámetro “hMenu”, le pasamos la variable utilizada en la función CreateMenu. El segundo y tercer parámetro funcionaran de acuerdo a lo que se le asigne. Para entender esto expliquemos algo: un elemento puede ser insertado por “POSICIÓN” o por su “IDENTIFICADOR DE MENU”. ¿Qué significa?. Vamos al grafico del menú básico que colocamos arriba:


Si ya hago esto: InsertMenuItem hBarra, 2, True, Prueba, suponiendo que el menú de la barra de menú esta en la variable hBarra, ¿Dónde se colocara el elemento prueba?. Como segundo parámetro indique la posición que me interesa, que en este caso es 2, si en el segundo parámetro esta la posición del elemento, entonces en el tercer parámetro se colocará TRUE, con esto se le indica a la función que en el segundo parámetro se encuentra es la POSICIÓN del elemento.

Es decir, “Salir” sería desplazado a la posición 3 ya que “Prueba” estaría en la posición 2.

En caso de que haya colocado: InsertMenuItem hBarra, ID_PRUEBA, False, Prueba, al poner false en el tercer parámetro implica que tenemos que colocar el ID del elemento en el segundo parámetro. ¿Pero, a donde se coloca el elemento?. Se coloca al final del menú, es decir, después del elemento “Salir” o lo que es lo mismo en la posición 3.

Así es como se manejan los parámetro 2 y 3 de esta función, como cuarto parámetro se le pasa la estructura del elemento a insertar.

En nuestro código lo usamos de esta manera:

    InsertMenuItem hMenu1, 0, True, Archivo

    InsertMenuItem hMenu1, 1, True, Edicion

    InsertMenuItem hMenu1, 2, True, Salir

Pueden darse cuenta que estoy ubicando los elementos por “posición”.

Ya creado los elementos, en el menú, lo que nos falta es asignarle el menú al formulario, y para eso es la función que viene a continuación.

SetMenu

Api: Declare Function SetMenu Lib "user32" (ByVal hwnd As Long, ByVal hMenu As Long) As Long

Esta función es la encargada de asignarle un menú a una ventana, especificada en el primer parámetro, y como segundo parámetro le pasamos el menú que queremos asignar.

DrawMenuBar

Api: Declare Function DrawMenuBar Lib "user32" (ByVal hwnd As Long) As Long

Esta función se encarga de “repintar” la barra de menú (menú bar) de una ventana, cuyo “hwnd” se especifica en el primer parámetro.

NOTA 

¿Cómo capturamos los mensajes de los menú?. Los mensajes que envían los menús, se encuentran especificado en la documentación oficial, en este proyecto nosotros capturamos los mensajes WM_MENUSELECT y WM_COMMAND, dejo claro que WM_COMMAND no es especifico del menú, ya que un botón, por ejemplo, también manda dicho mensaje, WM_COMMAND se origina cuando hacemos clic en un elemento de acción del menú, es decir, un elemento que abre otro submenú no manda dicho mensaje, solo los elementos de acción “command” mandan el mensaje.

El WM_MENUSELECT es enviado cuando se selecciona un elemento de acción del menú. OJO seleccionar no es darle clic al elemento, es marcarlo nada mas, es decir, tener el puntero del mouse encima de él. En nuestro gestor de mensajes usamos el WM_MENUSELECT de la siguiente manera:

     If mensaje = WM_MENUSELECT Then

        Select Case loword(wParam)

        Case ID_NUEVO

            frmMenu.Label1 = "ESTAS EN NUEVO"

        Case ID_SELECCION

            frmMenu.Label1 = "ESTAS EN SELECCCION"

        Case ID_SELECCION2

            frmMenu.Label1 = "ESTAS EN SELECCCION 2"

        End Select

    End If

¿Cómo saber que elemento esta seleccionado?. La parte baja de memoria de la variable “wParam” contiene dicho ID. Este ID es el valor que se le pasa al estructura del elemento en su parámetro “wID”.

¿Qué es parte baja de la memoria?.  “wParam” es una variable de 32 Bits, que es lo equivalente a 16 bits + 16 bits. Bueno existen dos funciones llamadas hiword y loword, que pueden tomar esos 16 bits, en este caso usamos loword(wParam), ya que la documentación oficial establece que en los 16 bits de la parte baja de la memoria de dicha variable se encuentra el valor del ID del elemento seleccionado. Y en la parte alta se encuentra una especie de “flag”, es decir, tiene un valor que dependiendo de dicho valor pueden saber si el elemento esta activado, desactivado, etc. Para ver dichos valores vayan a la ayuda de MSDN y busquen dicho mensaje.

¿En el mensaje WM_COMMAND, como saber el elemento seleccionado?. En este caso también la parte baja de la memoria que ocupa “wParam” contiene dicho valor. ¿Pero si en el código veo en el Select a wParam y no loword(wParam)?.

Ejm:

If mensaje = WM_COMMAND Then

        Select Case wParam

Lo que sucede es que en los menús la parte alta de dicha memoria es 0, por lo que da lo mismo usar wParam o loword(wParam). ¿Siempre es así?, cuando trabajamos con menús SI!!, pero con otros controles NO!!, recuerden que el mensaje WM_COMMAND no es exclusivo de los menús, sino es usado por otros controles. Si usamos un botón la parte alta de “wParam” no es 0, sino que posee un valor. De todas maneras en la ayuda pueden documentarse un poco mas.

GetMenuItemInfo

Api: Declare Function GetMenuItemInfo Lib "user32" Alias "GetMenuItemInfoA" (ByVal hMenu As Long, ByVal un As Long, ByVal b As Boolean, lpMenuItemInfo As MENUITEMINFO) As Long

Con esta función tomamos, o retornamos información de un elemento. Los parámetros son iguales a la función encargada de insertar dichos elementos. Por lo que me ahorro repetir la explicación. Lo que voy a explicar es como usamos dicha función. En el menú “hMenu1”, tenemos que dos elementos son del tipo CHECK, es decir, aquellos que pintan o una imagen en el lado izquierdo o un circulito negro al lado del texto.

Si usaran el editor de menú de VB, VB se encarga automáticamente de marcar y desmarcar dicho elemento, pero como creamos el menú a “mano” nosotros somos los encargados de marcar y desmarcar dicho elemento. ¿Cómo lo hacemos?. Capturamos el mensaje WM_COMMAND. Para saber si esta marcado o desmarcado obtenemos dicha información con la función  GetMenuItemInfo  ¿Cómo?. Veamos el ejemplo:

            inf.cbSize = Len(inf)

            inf.fMask = MIIM_STATE
            GetMenuItemInfo hMenu2, ID_SELECCION, False, inf

En “fMask” especificamos que parámetro deseamos recuperar, en nuestro caso es MIIM_STATE, es decir, el estado actual del elemento, por consiguiente, después de ejecutar la función GetMenuItemInfo, el parámetro de la estructura “fState” contendrá el valor actual:

            If inf.fState = MFS_CHECKED Then

                inf.fState = MFS_UNCHECKED

            Else

                inf.fState = MFS_CHECKED

            End If

Verifico dicho estado, y si esta seleccionado, lo deselecciono o viceversa. Ya cambiada la información, ahora hay que asignársela al elemento en cuestión.

SetMenuItemInfo

Api: Declare Function SetMenuItemInfo Lib "user32" Alias "SetMenuItemInfoA" (ByVal hMenu As Long, ByVal un As Long, ByVal bool As Boolean, lpcMenuItemInfo As MENUITEMINFO) As Long

Y esta función se encarga de ello, cuando queremos modificar información de un elemento ya creado tenemos que usar esta función. Los 4 parámetros ya fueron explicado. En nuestro ejemplo lo tenemos de esta manera:

            SetMenuItemInfo hMenu2, ID_SELECCION, False, inf

Pero dense cuenta, que la ubicación de dicho elemento no lo hacemos por “POSICION” sino usando el identificador del menú, ya que así es mas cómodo.

Por esa razón al presionar dicho elemento, el se marca si estaba desmarcado, o se desmarca si estaba marcado., La diferencia entre los dos elementos de selección, es que uno usa el circulito negro y el otro usa imágenes para indicar la selección, y deselección.

Es evidente que va ser muy raro, crear un menú de esta manera, pero es importante saber que funciones participan en dicha creación, por que ahora vamos a ver como crear un menú que no puede ser creado con el editor de VB, que son los menús “owner draw”.

MENU “OWNER DRAW”

Lo que vamos hacer en este proyecto es algo que no podemos hacer con el editor de VB, en primer lugar estarán preguntándose ¿Qué es OWNER DRAW?. Owner Draw implica que el encargado de dibujar el menú eres tu!.  Windows se lava las manos y deja de parte del programador dibujar dicho menú. Ustedes dirán ¡que fastidioso!, ciertamente lo es, pero se pueden lograr efectos espectaculares que no son posibles si dejamos que Windows se encargué del dibujo.

Por ejemplo, imaginen que tienen un menú cuyo elemento se llama “COLORES”, al seleccionar este elemento, usando la forma tradicional, mostraríamos un submenú con los nombres de los colores que vamos a utilizar, y cada ves que seleccionemos dicho nombre el fondo de nuestro formulario se pinta de dicho color. Pero no sería sorprendente que en ves de mostrar el nombre de los colores, el menú mostrara el “color”, es decir, que si el primer elemento es azul, no muestre el nombre “Azul” sino que dicho elemento se dibuje de azul. Dicho efecto sería imposible con el editor de menú tradicional y hay que usar las funciones de Windows y SubClassing para lograr dicho efecto.

Para este proyecto es necesario, crear un menú con el editor de VB, que va a tener un solo elemento en la barra de menú, cuyo nombre le pueden poner “menú Grafico”, a este menú le van a poner un elemento cuyo caption sea “Colores”, es decir, cuando se le de clic a “menú Grafico” se abrirá un submenú que tiene el elemento “Colores”, la parte de “name” les dejo a ustedes que pongan lo que quieran. Al formulario le ponen el nombre de “frmmenudraw”

Luego en un modulo pegan este código:

Option Explicit

Public Const MIIM_DATA = &H20

Public Const MIIM_FTYPE = &H100

Public Const MIIM_ID = &H2

Public Const MIIM_STATE = &H1

Public Const MIIM_SUBMENU = &H4

Public Const MIIM_TYPE = &H10

Public Const MFT_OWNERDRAW = &H100&

Public Const MFS_ENABLED = &H0&

Public Const IMAGE_ICON = 1

Public Const IMAGE_BITMAP = 0

Public Const LR_LOADFROMFILE = &H10

Public Type Rect

    Left As Long

    Top As Long

    Right As Long

    Bottom As Long

End Type

Public Type MENUITEMINFO

    cbSize As Long

    fMask As Long

    fType As Long

    fState As Long

    wID As Long

    hSubMenu As Long

    hbmpChecked As Long

    hbmpUnchecked As Long

    dwItemData As Long

    dwTypeData As String

    cch As Long

End Type

Public Type MEASUREITEMSTRUCT

    CtlType As Long

    CtlID As Long

    itemID As Long

    itemWidth As Long

    itemHeight As Long

    itemData As Long

End Type

Public Type DRAWITEMSTRUCT

    CtlType As Long

    CtlID As Long

    itemID As Long

    itemAction As Long

    itemState As Long

    hwndItem As Long

    hdc As Long

    rcItem As Rect

    itemData As Long

End Type

Declare Function CreateMenu Lib "user32" () As Long

Declare Function InsertMenuItem Lib "user32" Alias "InsertMenuItemA" (ByVal hMenu As Long, ByVal un As Long, ByVal bool As Boolean, lpcMenuItemInfo As MENUITEMINFO) As Long

Declare Function DestroyMenu Lib "user32" (ByVal hMenu As Long) As Long

Declare Function GetMenu Lib "user32" (ByVal hwnd As Long) As Long

Declare Function GetSubMenu Lib "user32" (ByVal hMenu As Long, ByVal nPos As Long) As Long

Declare Function GetMenuItemInfo Lib "user32" Alias "GetMenuItemInfoA" (ByVal hMenu As Long, ByVal un As Long, ByVal b As Boolean, lpMenuItemInfo As MENUITEMINFO) As Long

Declare Function SetMenuItemInfo Lib "user32" Alias "SetMenuItemInfoA" (ByVal hMenu As Long, ByVal un As Long, ByVal bool As Boolean, lpcMenuItemInfo As MENUITEMINFO) As Long

Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Declare Function FillRect Lib "user32" (ByVal hdc As Long, lpRect As Rect, ByVal hBrush As Long) As Long

Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lparam As Long) As Long

Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal un1 As Long, ByVal n1 As Long, ByVal n2 As Long, ByVal un2 As Long) As Long

Declare Sub MoveMemory Lib "kernel32" Alias "RtlMoveMemory" (dest As Any, ByVal Source As Long, ByVal length As Long)

Declare Function SetBkMode Lib "gdi32" (ByVal hdc As Long, ByVal nBkMode As Long) As Long

Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Declare Function loword Lib "TLBINF32" (ByVal DWord As Long) As Integer

Declare Function hiword Lib "TLBINF32" (ByVal DWord As Long) As Integer

Public Const GWL_WNDPROC = (-4)

Const WM_COMMAND = &H111

Const WM_MENUSELECT = &H11F

Const WM_DRAWITEM = &H2B

Const WM_MEASUREITEM = &H2C

Const ODA_DRAWENTIRE = &H1

Const ODA_SELECT = &H2

Const ODS_SELECTED = &H1

Const TRANSPARENT = 1

Const OPAQUE = 2

Dim ViejoGestor As Long

Public entro As Boolean

Public Const ID_COLOR_AMARILLO = 5

Public Const ID_COLOR_AZUL = 6

Public Const ID_COLOR_ROJO = 7

Public Const ID_MENU1 = 1

Public Const ID_MENU2 = 2

Public hMenu1 As Long

Public hMenu2 As Long

Public Amarillo As MENUITEMINFO

Public Azul As MENUITEMINFO

Public Rojo As MENUITEMINFO

Public Archivo As MENUITEMINFO

Public inf As MENUITEMINFO

Public item As MEASUREITEMSTRUCT

Public draw As DRAWITEMSTRUCT

Public r As Rect

Const height_menu = 30

Const width_menu = 60

Dim brocha As Long

Public Sub nuevoGestor(hwnd As Long)

    ViejoGestor = SetWindowLong(hwnd, GWL_WNDPROC, AddressOf GestorMensaje)

End Sub

Function GestorMensaje(ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByVal lparam As Long) As Long

    Select Case mensaje

    Case WM_MEASUREITEM

        Dim a As Long

        a = CallWindowProc(AddressOf GestorMensajeI, hwnd, mensaje, wParam, lparam)

    Case WM_DRAWITEM

        MoveMemory draw, lparam, Len(draw)

        Select Case draw.itemID

        Case ID_COLOR_AMARILLO

            Select Case draw.itemAction

            Case ODA_DRAWENTIRE

                FillRect draw.hdc, draw.rcItem, CreateSolidBrush(RGB(255, 255, 0))

            Case ODA_SELECT

                If draw.itemState And  ODS_SELECTED Then

                    brocha = CreateSolidBrush(RGB(255, 255, 0))

                    SelectObject draw.hdc, brocha

                    Rectangle draw.hdc, draw.rcItem.Left, draw.rcItem.Top, draw.rcItem.Right, draw.rcItem.Bottom

                    DeleteObject brocha

                Else

                    FillRect draw.hdc, draw.rcItem, CreateSolidBrush(RGB(255, 255, 0))

                End If

            End Select

        Case ID_COLOR_AZUL

            Select Case draw.itemAction

            Case ODA_DRAWENTIRE

                FillRect draw.hdc, draw.rcItem, CreateSolidBrush(RGB(0, 0, 255))

            Case ODA_SELECT

                If draw.itemState And ODS_SELECTED Then

                    brocha = CreateSolidBrush(RGB(0, 0, 255))

                    SelectObject draw.hdc, brocha

                    Rectangle draw.hdc, draw.rcItem.Left, draw.rcItem.Top, draw.rcItem.Right, draw.rcItem.Bottom

                    DeleteObject brocha

                Else

                    FillRect draw.hdc, draw.rcItem, CreateSolidBrush(RGB(0, 0, 255))

                End If

            End Select

        Case ID_COLOR_ROJO

            Select Case draw.itemAction

            Case ODA_DRAWENTIRE

                FillRect draw.hdc, draw.rcItem, CreateSolidBrush(RGB(255, 0, 0))

            Case ODA_SELECT

                If draw.itemState And ODS_SELECTED Then

                    brocha = CreateSolidBrush(RGB(255, 0, 0))

                    SelectObject draw.hdc, brocha

                    Rectangle draw.hdc, draw.rcItem.Left, draw.rcItem.Top, draw.rcItem.Right, draw.rcItem.Bottom

                    DeleteObject brocha

                Else

                    FillRect draw.hdc, draw.rcItem, CreateSolidBrush(RGB(255, 0, 0))

                End If

            End Select

        End Select

    Case WM_COMMAND

        Select Case wParam

            Case ID_COLOR_AMARILLO

                frmmenudraw.PintarFondo RGB(255, 255, 0)

            Case ID_COLOR_AZUL

                frmmenudraw.PintarFondo RGB(0, 0, 255)

            Case ID_COLOR_ROJO

                frmmenudraw.PintarFondo RGB(255, 0, 0)

        End Select

    End Select

    GestorMensaje = CallWindowProc(ViejoGestor, hwnd, mensaje, wParam, lparam)

End Function

Function GestorMensajeI(ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByRef item As MEASUREITEMSTRUCT) As Long

    Select Case mensaje

    Case WM_MEASUREITEM

        item.itemHeight = height_menu

        item.itemWidth = width_menu

    End Select

End Function

En el formulario:

Option Explicit

Private Sub Form_Load()

    nuevoGestor Me.hwnd

    CreaMenuSubMenu

End Sub

Sub InicializarEstructurasSubMenu()

    Amarillo.cbSize = Len(Amarillo)

    Amarillo.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Amarillo.fType = MFT_OWNERDRAW

    Amarillo.fState = MFS_ENABLED

    Amarillo.wID = ID_COLOR_AMARILLO

    Amarillo.hSubMenu = 0

    Amarillo.dwItemData = 0

    Amarillo.dwTypeData = ""

    Amarillo.cch = 0

    Amarillo.hbmpChecked = 0

    Amarillo.hbmpUnchecked = 0

    Rojo.cbSize = Len(Rojo)

    Rojo.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Rojo.fType = MFT_OWNERDRAW

    Rojo.fState = MFS_ENABLED

    Rojo.wID = ID_COLOR_ROJO

    Rojo.hSubMenu = 0

    Rojo.dwItemData = 0

    Rojo.dwTypeData = ""

    Rojo.cch = 0

    Rojo.hbmpChecked = 0

    Rojo.hbmpUnchecked = 0

    Azul.cbSize = Len(Azul)

    Azul.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Azul.fType = MFT_OWNERDRAW

    Azul.fState = MFS_ENABLED

    Azul.wID = ID_COLOR_AZUL

    Azul.hSubMenu = 0

    Azul.dwItemData = 0

    Azul.dwTypeData = ""

    Azul.cch = 0

    Azul.hbmpChecked = 0

    Azul.hbmpUnchecked = 0

End Sub

Sub CreaMenuSubMenu()

    hMenu1 = GetSubMenu(GetMenu(Me.hwnd), 0)

    hMenu2 = CreateMenu()

    InicializarEstructurasSubMenu

    InsertMenuItem hMenu2, 0, True, Amarillo

    InsertMenuItem hMenu2, 1, True, Rojo

    InsertMenuItem hMenu2, 2, True, Azul

    Archivo.cbSize = Len(Archivo)

    Archivo.fMask = MIIM_SUBMENU

    Archivo.hSubMenu = hMenu2

    SetMenuItemInfo hMenu1, 0, True, Archivo

End Sub

Sub PintarFondo(color As Long)

    Me.BackColor = color

End Sub

Private Sub Form_Unload(Cancel As Integer)

    DestroyMenu hMenu2

End Sub
Cuando arrancan el proyecto, se meten por “Menú Grafico”, verán que el elemento “Colores” abre un submenú, pero este submenú son 3 colores, que al ser seleccionados pintan el fondo del formulario de dicho color. ¡Sorprendente!. En este proyecto no hay muchas funciones nuevas que explicar, solamente hay dos sencillas, pero voy a concentrar la explicación al como funciona el “protocolo para crear un menú Owner Draw”.

¿Qué es esto de protocolo?. Me refiero a los pasos a seguir para lograr dicho efecto. Estos pasos están llenos de mucho detalles así que presten mucha atención, ya que explicare dicho código paso a paso.

Empecemos entonces en el load:

Private Sub Form_Load()

    nuevoGestor Me.hwnd ‘ Cambiamos el gestor de mensajes, ya deberían de estar familiarizados con esto.

    CreaMenuSubMenu ‘ llamamos a la función en donde creamos el submenú que va a contener dichos colores.

End Sub
Vamonos a “CreaMenuSubMenu”:

    hMenu1 = GetSubMenu(GetMenu(Me.hwnd), 0) ‘ Obtenemos el submenú que creamos con el editor de VB.

    hMenu2 = CreateMenu() ‘ Inicializamos nuestro nuevo menú.

    InicializarEstructurasSubMenu ‘ Configuramos las estructuras que contienen la configuración de los elementos que conformaran dicho submenú.

    InsertMenuItem hMenu2, 0, True, Amarillo ‘ Insertamos los elementos

    InsertMenuItem hMenu2, 1, True, Rojo

    InsertMenuItem hMenu2, 2, True, Azul

    Archivo.cbSize = Len(Archivo) ‘ Tomamos la longitud de la estructura llamada “Archivo”.

    Archivo.fMask = MIIM_SUBMENU ‘ Indico que voy a modificar el submenú del elemento.        

    Archivo.hSubMenu = hMenu2 ‘ Coloco el menú o submenú que mostrara el elemento del menú.

    SetMenuItemInfo hMenu1, 0, True, Archivo ‘ A hMenu1 (el menú que contiene el elemento  “Colores” que creamos con en el editor), a dicho elemento de posición “0”, le asigno la estructura “Archivo”, lo que hicimos aquí fue cambiar los atributos del elemento “Colores”, el atributo que cambiamos fue de que ahora dicho elemento no es un elemento de acción, sino que es un elemento que abre un submenú. Este submenú es hMenu2.

GetMenu

Api: Declare Function GetMenu Lib "user32" (ByVal hwnd As Long) As Long

Esta función es utilizada para tomar la referencia del menú de una ventana, dicha ventana es indicada en el “hwnd” que se le pasa como primer parámetro.

GetSubMenu

Api: Declare Function GetSubMenu Lib "user32" (ByVal hMenu As Long, ByVal nPos As Long) As Long

Con esta función obtenemos un “submenú” o menú despejable de un menú. Como primer parámetro le pasamos el menú original “hMenu”, y como segundo parámetro le pasamos la posición del submenú que queremos capturar.

TODO CONFUSO!!. Veamos el ejemplo de arriba:

    hMenu1 = GetSubMenu(GetMenu(Me.hwnd), 0) 
¿Que tiene “hMenu1”? El tiene el submenú que contiene al elemento “Colores” ¿Por qué?. A la función GetSubMenu le pasamos como primer parámetro nuestro menú principal, GetMenu(Me.hwnd), con esta línea en azul estamos retornando el menú donde se encuentra el elemento “Menú Grafico”, ya obtenida la barra de menú, el paso que sigue es obtener la referencia del submenú que abre “Menú Grafico”, para ello usamos GetSubMenu, le pasamos el menú donde sabemos que esta dicho submenú, y como segundo parámetro pasamos la posición donde se encuentra, en este caso la posición es 0, ya que el submenú que contiene “Colores” se encuentra en la posición 0. 

En pocas palabras “hMenu1”, contiene el submenú o menú, que contiene el elemento “Colores”.

Con la función InicializarEstructurasSubMenu configuramos los elementos que contendrá el submenú que abrirá el elemento “Colores”. Mostremos uno solo y la explicación es valido para los demás:

    Amarillo.cbSize = Len(Amarillo)

    Amarillo.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Amarillo.fType = MFT_OWNERDRAW  ‘ Con esto indicamos que vamos a ser nosotros los encargados de dibujar dicho elemento.

    Amarillo.fState = MFS_ENABLED  ‘ Indicamos que esta activo

    Amarillo.wID = ID_COLOR_AMARILLO ‘ Le configuramos un ID único que vamos a usar cuando recibamos los mensajes.

    Amarillo.hSubMenu = 0

    Amarillo.dwItemData = 0

    Amarillo.dwTypeData = ""

    Amarillo.cch = 0

    Amarillo.hbmpChecked = 0

    Amarillo.hbmpUnchecked = 0
En el momento que colocamos esta constante MFT_OWNERDRAW  ya nuestra ventana sufre cambios, en el sentido que ahora va a recibir 2 mensajes, que antiguamente no recibía. Dichos mensajes son:  WM_MEASUREITEM y WM_DRAWITEM, para poder usar dichos mensajes tuve que recurrir a ciertas trampitas, así que presten mucha atención a lo que explicare a continuación.

WM_MEASUREITEM

Este mensaje se genera cuando se crea el menú o submenú (owner draw) por primera vez, y las veces que se genera dependerá de la cantidad de elementos que tenga el submenú, que quiero decir, si tienen un submenú con 5 elementos este mensaje se genera 5 veces cuando se despliega el menú por primera vez. ¿Qué hacemos aquí?. Normalmente lo que se hace es determinar el ancho y alto de dicho elemento, y hacer ciertas configuraciones preliminares a cada elemento antes de ser creado el submenú, en nuestro ejemplo, lo único que hacemos es cambiar el ancho y alto del elemento.

Cuando se genera dicho mensaje la variable “wParam” contiene 0, siempre y cuando sea un menú. En caso de que sea un ComboBox, Listbox u otro control, “wParam” contiene el valor que identifica a dicho control.

NOTA

¿ComboBox, ListBox u otro control?. Esta técnica de “Owner Draw” no es exclusivo de los menús, los ComboBox, Listbox, button, etc, también tienen la opción de personalizar los elementos que dichos controles muestran, se trabaja prácticamente de la misma manera, por lo que voy a explicar como se usa en los menús, y en caso de estar interesados de hacerlo en dichos controles, entonces vean la ayuda de MSDN. Pero si entienden esta parte, les será fácil llevar esta técnica de Owner Draw a dichos controles.

La variable interesante es “lParam”, esta variable contiene un puntero a una estructura del tipo MEASUREITEMSTRUCT. Esta estructura la presento a continuación:

Public Type MEASUREITEMSTRUCT

    CtlType As Long

    CtlID As Long

    itemID As Long

    itemWidth As Long

    itemHeight As Long

    itemData As Long

End Type

CtlType: identifica el tipo de control, puede tener estos valores:

Private Const ODT_BUTTON = 4

Private Const ODT_COMBOBOX = 3

Private Const ODT_LISTBOX = 2

Private Const ODT_LISTVIEW = 102

Private Const ODT_MENU = 1

Creo que en las constantes es fácil ver que control identifica cada una.

CtlID: Especifica el identificador de dicho control. En los menús no se usa.

ItemID: Especifica el identificador del elemento de un menú. En caso de ser un ComboBox o Listbox el parámetro contiene la posición del elemento.

ItemWidth: Determina el ancho del elemento.

ItemHeight: Determina el alto del elemento.

ItemData: Especifica una información adicional de 32 Bits, asociado con un elemento del menú. En caso de ser un ComboBox y Listbox, y dependiendo los estilos configurados en dichos controles tendrán un valor asignado. Para ver que valores pueden ser, vean en la ayuda de MSDN.

En los menús normalmente lo que interesa son los parámetros ItemWidth e ItemHeight, pero cambiar dichos valores es todo un reto, y no es tan fácil como lo imaginan ¿Por qué?. 

En primer lugar “lParam” es una variable de tipo “LONG”, es decir, que no se puede hacer esto:

LParam.ItemWidth = ancho

LParam.ItemHeight = alto

Para entender “el como lo hice”, voy a explicar lo que yo intente hacer al principio, que no sirvió y luego explicare “el por que” lo que hice después funciono.

Antes de eso, vamos a explicar una función que sirve para manipular información en la memoria.

MoveMemory

Api: Declare Sub MoveMemory Lib "kernel32" Alias "RtlMoveMemory" (dest As Any, ByVal Source As Long, ByVal length As Long)

Esta función mueve la información de un lugar de la memoria a otro lugar. Como primer parámetro pasamos la variable que recibirá dicha información, como segundo parámetro pasamos la variable que contiene la información a mover, y como tercer parámetro, pasamos la cantidad de bytes que queremos copiar.

Lo primero que intente hacer fue lo siguiente:

    Case WM_MEASUREITEM

MoveMemory item, lparam, Len(item)

item.itemHeight = height_menu

             item.itemWidth = width_menu

¿Que hiciste aquí Eduardo? Vean la siguiente grafica:

Memoria





“lParam” como dije es un puntero ¿Que es esto?. Es la dirección de memoria, de donde se encuentra la información de la estructura, pero esa estructura como explique arriba no la puedo acceder por “lParam” directamente ya que “lParam” es una variable de tipo “long”,  cuando se ejecuta esta línea:

MoveMemory item, lparam, Len(item)

La información de “lParam”, viaja y se mueve a la ubicación en memoria de la variable “item”, a partir de este momento en la memoria hay 2 bloques de información iguales ¿Ven por que?. Claro “lParam” sigue teniendo la información de antes, pero la moví a la variable “item”, el “mover” por las pruebas que hice no implica que “lparam” pierde la información, tan sencillamente mueve o superpone la información que había en “item” que originalmente no tenia nada, es decir, los parámetros tenían 0.

Aplicado esta instrucción, ya podía acceder a la información de la variable “item”, es decir, esta dos líneas eran validas:

item.itemHeight = height_menu

             item.itemWidth = width_menu

Cabe mencionar que Windows tiene un tamaño por defecto, es decir, ya ItemWidth e ItemHeight tienen un valor. Pero lo mencionado arriba no funcionaba, no importa el tamaño que colocara, los elementos del menú siempre aparecían con el mismo tamaño, al principio no sabia que pasaba, pero veamos de nuevo el grafico:




Es decir, en un bloque de memoria estaba la nueva altura y ancho que queríamos, y en el otro bloque estaban la altura y ancho por defecto. Inmediatamente me pregunte ¿Como Windows sabrá que se modifican los valores?. Y pensé!!, Windows debe de estar esperando que se modifique la estructura que se encuentra en la dirección de “lParam”, evidentemente no funcionaba, por que Windows no esta al tanto de saber que yo moví el bloque de memoria y que la información actual la posee la variable “item”. Windows espera que yo modifique el bloque de memoria que referencia “lParam”.

El reto ahora quedaba, en como modificar el bloque que referenciaba “lParam”, ustedes dirán regresas el cambio:

MoveMemory lParam, item, Len(item)

ERROR!!, ¿por qué? como segundo parámetro la función MoveWindow NO ACEPTA estructuras sino variables del tipo LONG, al principio lo acepto por que como primer parámetro se puede pasar cualquier cosa, ya que la variable era del tipo “Any”. Segundo “lParam” es una variable de 4 bytes, y la estructura MEASUREITEMSTRUCT es de 24 Bytes, recuerden que “lparam” contiene es la dirección de memoria donde empieza la estructura.

¿Qué hacemos entonces? Creamos otra función que funcione como “oído” en este caso GestorMensajeI, pero este oído es especial:

Function GestorMensajeI(ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByRef item As MEASUREITEMSTRUCT) As Long

Se supone, que en todo gestor de mensajes “normal” en el último parámetro debería de estar la variable “lParam”, pero le hice una modificación al gestor, por lo que ahora recibe una “referencia” del tipo MEASUREITEMSTRUCT.
Cuando el gestor de mensajes principal recibe el mensaje, llamamos a nuestro “oído” especial:

        Dim a As Long

        a = CallWindowProc(AddressOf GestorMensajeI, hwnd, mensaje, wParam, lparam)

Ustedes se preguntaran ¿Por que funciona?. Veamos la grafica:








Por consiguiente, cuando Windows toma el bloque de memoria referenciado por “lParam” esta tomando la estructura con la nueva información modificada. 

¿Es así tan complicado esto?. Esta parte NO!, es complicado en VB ya que esta lenguaje carece de punteros, en C se soluciona con una simple línea:

lpmis = (LPMEASUREITEMSTRUCT) lParam
Pero como VB carece de este tipo de conversiones que posee C, y tenemos que usar trucos para solucionar el problema, yo conseguí solucionarlo de esta manera, si un experto en memoria logra hacerlo de manera mas fácil, le agradecería que me dijera ¿cómo?.

WM_DRAWITEM

Este mensaje es enviado cuando se le cambia la apariencia visual a un ComboBox, ListBox, menú, button, etc. Es decir, es en este mensaje es donde dibujamos lo que queremos mostrar. Al igual que WM_MEASUREITEM, “wParam” contiene el identificador del control que genero el mensaje, y  “lParam” tiene un puntero a una estructura del tipo DRAWITEMSTRUCT.

Public Type DRAWITEMSTRUCT

    CtlType As Long

    CtlID As Long

    itemID As Long

    itemAction As Long

    itemState As Long

    hwndItem As Long

    hdc As Long

    rcItem As Rect

    itemData As Long

End Type

CtlType : identifica el tipo de control, puede tener los mismos valores especificados en el “CtlType” de la estructura MEASUREITEMSTRUCT.

CtlId: Especifica el identificador de dicho control. En los menús no se usa.

ItemID: Especifica el identificador del elemento de un menú. En caso de ser un ComboBox o Listbox el parámetro contiene la posición del elemento.

ItemAction: Especifica la acción que se ejecuta en dicho control. Puede tomar los siguientes valores:

Private Const ODA_DRAWENTIRE = &H1 : Este parámetro determina que el control debe de ser pintado por completo.

Private Const ODA_FOCUS = &H4 : El control ha perdido o ganado el foco.

Private Const ODA_SELECT = &H2  : El status de selección del elemento de dicho control ha cambiado.

ItemState: Especifica el estado del elemento. Con ello tomamos la decisión de ¿Qué pintar?. Puede tener los siguientes valores:

Private Const ODS_CHECKED = &H8 : El elemento del menú se encuentra en el estado “check” o “marcado”.

Private Const ODS_DEFAULT = &H20 : El elemento seleccionado es el que se encuentra configurado como elemento “por defecto”.

Private Const ODS_DISABLED = &H4 : El elemento se encuentra desactivado.

Private Const ODS_FOCUS = &H10 : El elemento tiene el foco del teclado.

Private Const ODS_GRAYED = &H2 : El elemento se encuentra configurado como desactivado. Solo para menús.

Private Const ODS_SELECTED = &H1 : El elemento se encuentra seleccionado

HwndItem: contiene el hwnd del control que contiene dichos elementos, en el caso de los menús, contiene el identificador de dicho menú.

Hdc: dispositivo de contexto, del control en donde dibujamos.

RcItem: Especifica el tamaño del rectángulo en donde podemos dibujar.

ItemData: Especifica una información adicional de 32 Bits, asociado con un elemento del menú. En caso de ser un ComboBox y Listbox, y dependiendo los estilos configurados en dichos controles tendrán un valor asignado. Para ver que valores puede ser vean en la ayuda de MSDN.

Estos parámetros, lo usamos para saber básicamente el estado del elemento activo, para así poder dibujar, recuerden que “owner draw” implica que tenemos que dibujar ¡TODO!, nosotros somos los encargado de programar lo que ve el usuario cuando un elemento tiene un estado especifico.

¿Cómo usamos este mensaje en el código?. Primero movemos el bloque de memoria de “lParam” a un bloque de memoria que ya esta configurado para recibir una estructura del tipo DRAWITEMSTRUCT.

Case WM_DRAWITEM

        MoveMemory draw, lparam, Len(draw)

¿Por que aquí, usamos MoveMemory y no la técnica empleada con MEASUREITEMSTRUCT?. Esta estructura es de carácter informativo, es decir, no vamos a cambiar nada, el problema con la estructura anterior es que teníamos que cambiar la altura y ancho a nuestra conveniencia, y esta técnica no servia. Pero como solo vamos a usar esta estructura de carácter informativo, el usar MoveMemory sirve a la perfección.

Select Case draw.itemID

Con esto verificamos que elemento esta pidiendo ser dibujado.

Case ID_COLOR_AMARILLO

            Select Case draw.itemAction

Poniendo el ejemplo que el elemento que solicita ser pintado es ID_COLOR_AMARILLO, procedemos a verificar que ACCIÓN se esta ejecutando en dicho elemento.

Case ODA_DRAWENTIRE

                FillRect draw.hdc, draw.rcItem, CreateSolidBrush(RGB(255, 255, 0))

Los elementos ejecutaran este mensaje ODA_DRAWENTIRE, cada vez que el Sub-Menú se muestre, es decir, cada vez que cerremos el menú y lo volvamos abrir, se ejecuta dicho mensaje, indicando que debe de ser pintado dicho elemento.

Case ODA_SELECT

                If draw.itemState And ODS_SELECTED Then

                    brocha = CreateSolidBrush(RGB(255, 255, 0))

                    SelectObject draw.hdc, brocha

                    Rectangle draw.hdc, draw.rcItem.Left, draw.rcItem.Top, draw.rcItem.Right, draw.rcItem.Bottom

                    DeleteObject brocha

                Else

                    FillRect draw.hdc, draw.rcItem, CreateSolidBrush(RGB(255, 255, 0))

                End If

Ya con el menú abierto, no se ejecuta mas ODA_DRAWENTIRE, y capturamos la acción ODA_SELECT, es decir, cuando pasamos el mouse encima de los elementos la acción que se ejecuta es ODA_SELECT. Verificamos el estado de dicho elemento: 
If draw.itemState And ODS_SELECTED Then

Si esta seleccionado, es decir, el puntero del mouse esta encima del elemento, pintamos un rectángulo negro alrededor, para indicarle al usuario que ese es el color que se encuentra seleccionado.

En caso contrario pintamos de nuevo el rectángulo, sin el marco negro, indicando que dicho elemento fue deseleccionado.

Cuando el mouse esta sobre un elemento y pasa a otro elemento, Windows manda dos veces el mensaje WM_DRAWITEM, el primero es para el elemento que fue deseleccionado y el segundo es para el elemento actualmente seleccionado.

En este ejemplo nada mas programe un solo estado que es ODS_SELECTED. Dependiendo de lo que haga su menú, o control, deberán de programar los otros diferentes estados que pueden tener dicho elemento.

¿Cómo sabias el tamaño del rectángulo a dibujar?. El parámetro rcItem, contiene el rectángulo del elemento a dibujar. OJO : “del elemento a dibujar”, en este caso este parámetro no tiene el tamaño de todo el hdc, sino nada mas tiene el tamaño y ubicación de la porción que le corresponde a dicho elemento. Estos cálculos Windows los hace automáticamente, gracias a que en el mensaje WM_MEASUREITEM nosotros configuramos el ancho y alto de los elementos.

Pueden ver que el trabajo de esta técnica “Owner Draw” sin importar donde la apliquemos, ya sea en un menú, comboBox o listbox, es un trabajo arduo y largo, y ustedes dirán ¿Vale la pena?. Por supuesto, imaginen por ejemplo, que tienen un menú donde el usuario, selecciona un tipo de letra, e imaginen que el texto del menú, se encuentre escrito con el tipo de fuente que se quiere seleccionar, o todos esos menús con texturas, imágenes, efectos especiales, que ya con esta información ustedes se encuentran en la capacidad de desarrollarlos sin ningún problema.

MENU DEL SISTEMA

En este proyecto vamos a manipular el menú del sistema que posee un formulario, me refiero al menú mostrado cuando se le da clic al icono del formulario, a este menú se le llama “menú del sistema” o “system menú”. Para este proyecto necesitan un formulario llamado “frmmenusistema” que tenga 3 botones y un modulo.

En el modulo pegan este código:

Option Explicit

Public Const MIIM_ID = &H2

Public Const MIIM_STATE = &H1

Public Const MIIM_TYPE = &H10

Public Const MFT_STRING = &H0&

Public Const MFS_ENABLED = &H0&

Public Type Rect

    Left As Long

    Top As Long

    Right As Long

    Bottom As Long

End Type

Public Type MENUITEMINFO

    cbSize As Long

    fMask As Long

    fType As Long

    fState As Long

    wID As Long

    hSubMenu As Long

    hbmpChecked As Long

    hbmpUnchecked As Long

    dwItemData As Long

    dwTypeData As String

    cch As Long

End Type

Declare Function InsertMenuItem Lib "user32" Alias "InsertMenuItemA" (ByVal hMenu As Long, ByVal un As Long, ByVal bool As Boolean, lpcMenuItemInfo As MENUITEMINFO) As Long

Declare Function DrawMenuBar Lib "user32" (ByVal hwnd As Long) As Long

Declare Function EnableMenuItem Lib "user32" (ByVal hMenu As Long, ByVal wIDEnableItem As Long, ByVal wEnable As Long) As Long

Declare Function GetMenuItemCount Lib "user32" (ByVal hMenu As Long) As Long

Declare Function GetSystemMenu Lib "user32" (ByVal hwnd As Long, ByVal bRevert As Long) As Long

Declare Function RemoveMenu Lib "user32" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long) As Long

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Public Const SC_CLOSE = 61536

Public Const SC_CONTEXTHELP = 61824

Public Const SC_DEFAULT = 61792

Public Const SC_HOTKEY = 61776

Public Const SC_HSCROLL = 61568

Public Const SC_KEYMENU = 61696

Public Const SC_MAXIMIZE = 61488

Public Const SC_MINIMIZE = 61472

Public Const SC_MONITORPOWER = 61808

Public Const SC_MOUSEMENU = 61584

Public Const SC_MOVE = 61456

Public Const SC_NEXTWINDOW = 61504

Public Const SC_PREVWINDOW = 61520

Public Const SC_RESTORE = 61728

Public Const SC_SCREENSAVE = 61760

Public Const SC_SIZE = 61440

Public Const SC_TASKLIST = 61744

Public Const SC_VSCROLL = 61552

Public Const MF_BYCOMMAND = &H0&

Public Const MF_BYPOSITION = &H400&

Public Const MF_DISABLED = &H2&

Public Const MF_ENABLED = &H0&

Public Const MF_GRAYED = &H1&

Public Const GWL_WNDPROC = (-4)

Const WM_COMMAND = &H111

Const WM_SYSCOMMAND = &H112

Dim ViejoGestor As Long

Public entro As Boolean

Public Const ID_NUEVO = 5

Public Nuevo As MENUITEMINFO

Public desactivar As Boolean

Public Sub nuevoGestor(hwnd As Long)

    ViejoGestor = SetWindowLong(hwnd, GWL_WNDPROC, AddressOf GestorMensaje)

End Sub

Public Function GestorMensaje(ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    Select Case mensaje

    Case WM_SYSCOMMAND

        Select Case wParam

        Case SC_CLOSE

            MsgBox "Se procedera a cerrar la aplicacion"

        Case SC_MAXIMIZE

            MsgBox "Se procedera a maximizar la aplicacion"

        Case SC_MINIMIZE

            MsgBox "Se procedera a inimizar la aplicacion"

        End Select

    End Select

    GestorMensaje = CallWindowProc(ViejoGestor, hwnd, mensaje, wParam, lParam)

End Function
En el formulario:

Option Explicit

Private Sub Command1_Click()

    InsertMenuItem GetSystemMenu(Me.hwnd, False), GetMenuItemCount(GetSystemMenu(Me.hwnd, False)) + 1, True, Nuevo

End Sub

Private Sub Command2_Click()

    GetSystemMenu Me.hwnd, True

    DrawMenuBar Me.hwnd

End Sub

Private Sub Command3_Click()

    RemoveMenu GetSystemMenu(Me.hwnd, False), SC_CLOSE, MF_BYCOMMAND

    DrawMenuBar Me.hwnd

End Sub

Private Sub Form_Load()

    Command1.Height = 510

    Command1.Left = 360

    Command1.Top = 630

    Command1.Width = 1995

    Command1.Caption = "Introducir Nuevo Item al menú del Sistema"

    Command2.Height = 510

    Command2.Left = 2520

    Command2.Top = 630

    Command2.Width = 1995

    Command2.Caption = "Restablecer el menu del sistema"

    Command3.Height = 510

    Command3.Left = 1395

    Command3.Top = 1350

    Command3.Width = 1995

    Command3.Caption = "Anular el boton Cerrar"

    Nuevo.cbSize = Len(Nuevo)

    Nuevo.fMask = MIIM_ID Or MIIM_STATE Or MIIM_TYPE

    Nuevo.fType = MFT_STRING

    Nuevo.fState = MFS_ENABLED

    Nuevo.wID = ID_NUEVO

    Nuevo.hSubMenu = 0

    Nuevo.dwItemData = 0

    Nuevo.dwTypeData = "&Nuevo"

    Nuevo.cch = Len("Nuevo")

    Nuevo.hbmpChecked = 0

    Nuevo.hbmpUnchecked = 0

    nuevoGestor Me.hwnd

End Sub
Presionen los botones, y luego desplieguen el menú del sistema para que vean dichos efectos.

GetSystemMenu

Api: Declare Function GetSystemMenu Lib "user32" (ByVal hwnd As Long, ByVal bRevert As Long) As Long

Función que retorna el identificador o handle del menú del sistema. Como primer parámetro se le pasa la ventana que contiene al menú, y como segundo parámetro le pasamos TRUE o FALSE, si le indicamos FALSE, la función retorna el identificador del menú, en caso de pasarle TRUE el menú regresa a su estado original y retorna 0.

Podemos ver su uso en el evento:

Private Sub Command1_Click()

    InsertMenuItem GetSystemMenu(Me.hwnd, False), GetMenuItemCount(GetSystemMenu(Me.hwnd, False)) + 1, True, Nuevo

End Sub
Como pueden ver se usan las funciones de manera directa, es decir, el valor que me retorna la función no la guardo en una variable sino voy directo a la función que necesita dicho valor.

En el otro evento:

Private Sub Command2_Click()

    GetSystemMenu Me.hwnd, True

    DrawMenuBar Me.hwnd

End Sub
Restablezco el menú a su estado original.

GetMenuItemCount

Api: Declare Function GetMenuItemCount Lib "user32" (ByVal hMenu As Long) As Long

Función que se encarga de enumerar cuantos elementos posee un menú. Como único parámetro posee el menú al cual queremos saber cuantos elementos tiene.

Su uso lo pueden ver en Command1_Click.
RemoveMenu

Api: Declare Function RemoveMenu Lib "user32" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long) As Long

Esta función se usa para remover un elemento de un menú, este menú viene especificado en el primer parámetro de la función. Como segundo parámetro se le coloca, la posición del elemento o el identificador de dicho elemento, para que la función sepa “que valor se paso”, se usa el tercer parámetro, el cual puede tener los siguientes valores:

Public Const MF_BYCOMMAND = &H0& : Indica que al elemento se le esta identificando por su “identificador”, es decir, identificador especificado en el segundo parámetro.

Public Const MF_BYPOSITION = &H400& : Indica que al elemento se le esta identificando por su posición en el menú, es decir, el segundo parámetro tiene el valor de dicha posición.

En nuestro proyecto uso el MF_BYCOMMAND:

Private Sub Command3_Click()

    RemoveMenu GetSystemMenu(Me.hwnd, False), SC_CLOSE, MF_BYCOMMAND

    DrawMenuBar Me.hwnd

End Sub

WM_SYSCOMMAND

Este es el mensaje que es enviado a la ventana cuando algún elemento del sistema o del área nocliente es activado.

El valor que posee “wParam” es el tipo de mensaje enviado por el sistema. Y “lParam” se divide en dos valores, la parte alta “hiword” posee la coordenada “Y” del puntero del mouse. Y la parte baja “loword” posee la coordenada “X” del puntero del mouse.

Algunos de los tipos de mensajes que Windows puede enviarles a través del mensaje WM_SYSCOMMAND son:

Public Const SC_CLOSE = 61536 : Indica que la ventana será cerrada, es decir, se presiono la “X” del formulario, o se selecciono cerrar en el menú del sistema

Public Const SC_HSCROLL = 61568 : Se esta usando el Scroll horizontal de la ventana

Public Const SC_MAXIMIZE = 61488 : Indica que se maximizara la ventana, es decir, se presiono en el botón maximizar.

Public Const SC_MINIMIZE = 61472 : Indica que se minimizara la ventana, es decir, se presiono en el botón minimizar.

Public Const SC_MONITORPOWER = 61808 : Se manda cuando se activa el “ahorro de energía” o “power safe” por parte del computador o Windows, en este caso “lparam” puede contener 2 valores:

1 = significa que el sistema esta activo en baja energía o “low power”

2 = significa que el sistema regresa a la normalidad, es decir, se desactiva el ahorrador de energía.

Public Const SC_MOVE = 61456 : Se manda este mensaje cuando la ventana se mueve.

Public Const SC_RESTORE = 61728 : Indica que se restaura la ventana, es decir, se presiono en el botón restaurar.

Public Const SC_SCREENSAVE = 61760 : Indica que el protector de pantalla va ser activado.

Public Const SC_SIZE = 61440 : Indica que las dimensiones de la ventana se están alterando.

Public Const SC_VSCROLL = 61552 : Se esta usando el Scroll vertical de la ventana.

En el proyecto, se usa las constantes SC_CLOSE, SC_MAXIMIZE y SC_MINIMIZE, es decir, cada vez que presionan los botones de la esquina superior derecha de la ventana, captura dicho clic con el mensaje WM_SYSCOMMAND, y muestro un mensaje antes de que la acción solicitada se ejecute.

MENU CONTEXTUALES

Los menús contextuales, son esos menús que aparecen en las ventanas, cuando se presionan el botón secundario del mouse. Por ejemplo, si presionan en este documento el botón secundario del mouse, verán un menú corto, con distintas opciones como “editar”, “copiar” , etc, estos menús son llamados “menús contextuales”.

Para este proyecto, se necesita un formulario, que tenga un botón y un menú con la siguiente estructura:

Archivo

     Nuevo

     Salir

Y luego pegan este código, en el formulario:

Private Type RECT

    Left As Long

    Top As Long

    Right As Long

    Bottom As Long

End Type

Private Declare Function TrackPopupMenu Lib "user32" (ByVal hMenu As Long, ByVal wFlags As Long, ByVal X As Long, ByVal Y As Long, ByVal nReserved As Long, ByVal hwnd As Long, lprc As RECT) As Long

Private Declare Function GetMenu Lib "user32" (ByVal hwnd As Long) As Long

Private Declare Function GetSubMenu Lib "user32" (ByVal hMenu As Long, ByVal nPos As Long) As Long

Private Declare Function GetWindowRect Lib "user32" (ByVal hwnd As Long, lpRect As RECT) As Long

Private Const TPM_BOTTOMALIGN = &H20&

Private Const TPM_CENTERALIGN = &H4&

Private Const TPM_LEFTALIGN = &H0&

Private Const TPM_RIGHTALIGN = &H8&

Private Const TPM_TOPALIGN = &H0&

Private Const TPM_VCENTERALIGN = &H10&

Private Const TPM_RIGHTBUTTON = &H2&

Private Const TPM_LEFTBUTTON = &H0&

Private Sub Command1_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)

    If Button = 2 Then

        Dim r As RECT

        GetWindowRect Command1.hwnd, r

        r.Left = r.Left + (X \ 15)

        r.Top = r.Top + (Y \ 15)

        'TrackPopupMenu GetSubMenu(GetMenu(hwnd), 0), TPM_RIGHTBUTTON, r.Left, r.Top, 0, Me.hwnd, r

        'TrackPopupMenu GetSubMenu(GetMenu(hwnd), 0), TPM_CENTERALIGN, r.Left, r.Top, 0, Me.hwnd, r

        'TrackPopupMenu GetSubMenu(GetMenu(hwnd), 0), TPM_VCENTERALIGN, r.Left, r.Top, 0, Me.hwnd, r

        'TrackPopupMenu GetSubMenu(GetMenu(hwnd), 0), TPM_BOTTOMALIGN, r.Left, r.Top, 0, Me.hwnd, r

        'TrackPopupMenu GetSubMenu(GetMenu(hwnd), 0), TPM_LEFTALIGN, r.Left, r.Top, 0, Me.hwnd, r

        'TrackPopupMenu GetSubMenu(GetMenu(hwnd), 0), TPM_RIGHTALIGN, r.Left, r.Top, 0, Me.hwnd, r

        TrackPopupMenu GetSubMenu(GetMenu(hwnd), 0), TPM_RIGHTALIGN, r.Left, r.Top, 0, Me.hwnd, r    End If

End Sub

Private Sub Form_Load()

    Command1.Caption = "Presiona con el botón secundario"

    Command1.Height = 510

    Command1.Left = 1215

    Command1.Top = 1260

    Command1.Width = 2130

End Sub
Cuando corran, presionen el botón secundario en el botón mostrado y verán que se muestra el submenú, que contiene el elemento “Archivo”.

Ustedes dirán, ¿Cuál es la gracia, de mostrar un menú que se muestra en el formulario? Ciertamente no tiene sentido, lo hice solo para enseñar el funcionamiento de la función encargada de mostrar dicho menú, en la “realidad” tienes 2 opciones, o creas el menú con “CreateMenu”, opción muy fastidiosa, ya que tenemos que crear elemento por elemento y capturar los mensajes de cada elemento por SubClassing. O la segunda opción, que la use en un proyecto personal, esta consistía en crear un formulario donde configuraba cada menú contextual con el editor de menú. Este formulario “JAMAS” lo mostraba el programa, es decir, no tenia nada, lo único que tenia eran los menús, cuando un botón necesitaba algún menú contextual, lo que hacia era, cargar el formulario, y tomaba con las funciones API’S el menú, sub menú, etc, y mostraba el menú, cuando el usuario da clic a un elemento, dicho evento lo capturo en el formulario que posee el menú y listo!!!, ¡rápido y sencillo!.

TrackPopupMenu

Api: Declare Function TrackPopupMenu Lib "user32" (ByVal hMenu As Long, ByVal wFlags As Long, ByVal X As Long, ByVal Y As Long, ByVal nReserved As Long, ByVal hwnd As Long, lprc As RECT) As Long

Esta es la función que se encarga de desplegar un menú, en una posición especifica. Como primer parámetro le pasamos el handle o identificador del menú, como segundo parámetro “wFlags” le podemos pasar unas constantes que se muestran a continuación:

Private Const TPM_BOTTOMALIGN = &H20& 

Private Const TPM_CENTERALIGN = &H4&

Private Const TPM_LEFTALIGN = &H0&

Private Const TPM_RIGHTALIGN = &H8&

Private Const TPM_TOPALIGN = &H0&

Private Const TPM_VCENTERALIGN = &H10&

Private Const TPM_RIGHTBUTTON = &H2&

Private Const TPM_LEFTBUTTON = &H0&

El efecto que tiene cada constante, véanlo en el proyecto, quitándole los comentarios a las líneas que contiene la función TrackPopupMenu.

Como segundo y tercer parámetro les pasamos las coordenadas X e Y donde será mostrado, dichas coordenadas son de “pantalla” o “screen”,  ¿Qué significa coordenada de pantallas o screen?. Todos sabemos que la configuración de video de nuestra pantalla, posee una resolución, por ejemplo, 800x600 píxeles, coordenadas en pantalla significa un X e Y dentro del rango de la resolución de pantalla.

Al parámetro “nReserved” se le pasa 0, al parámetro “hwnd” se le pasa el “hwnd” de la ventana que contiene dicho menú. Y el último parámetro es “lprc” que es una variable de tipo RECT, la cual no es tomada en cuenta, es decir, no importa que valor tenga Windows hace caso omiso a este parámetro.

GetWindowRect

Api: Declare Function GetWindowRect Lib "user32" (ByVal hwnd As Long, lpRect As RECT) As Long

Esta función lo que hace es tomar las coordenadas de “pantalla” de una ventana. Como primer parámetro le pasamos el “hwnd” de la ventana a la cual queremos saber su ubicación, y como segundo parámetro le pasamos una variable del tipo RECT en donde se almacenaran dichas coordenadas.

En nuestro proyecto la usamos para conocer las coordenadas de pantalla del botón:

        GetWindowRect Command1.hwnd, r

        r.Left = r.Left + (X \ 15)

        r.Top = r.Top + (Y \ 15)

Es decir, luego a esas coordenadas le sumamos, el lugar donde se presiono el botón secundario del mouse, divido entre 15 ya que recuerden que estas coordenadas están en Twips, y 1 Twip equivale a 15 píxeles, es esa la razón del porque el menú se abre justo en el puntero del mouse.

Capítulo  VII.  Teclado, Mouse y Multimedia
Básicamente, todos sabemos lo que hace el teclado y el mouse, estos 2 dispositivos nos permiten comunicarnos con el computador, es lo que mas usamos para interactuar con el. El teclado nos sirve para “escribir”, y el mouse básicamente nos sirve para manipular objetos en la computadora de manera mas sencilla, ya que manipular dichos elementos con el teclado harian muy difícil la interacción entre el usuario y la computadora.

¿Cómo funciona básicamente estos dos dispositivos?. Empecemos con el teclado, en este dispositivo, debajo de las teclas que ustedes presionan, presenta dispositivos electrónicos, cada vez que presionan una tecla un “scan code” es generado, dicho “scan code”, lo recibe Windows, y ese codigo lo convierte en otro codigo llamado codigo virtual “virtual code”, ¿Por qué Windows hace eso?. Windows lo hace debido a que un “scan code” puede representar diferentes letras dependiendo de la configuración regional del teclado. ¿Cómo?. Veamos un ejemplo sencillo, ustedes pueden tener un teclado, cuyas teclas son del idioma “ingles”, esto significa que por ningun lado ven la tecla “ñ”, pero en Windows pueden configurar el teclado como “español”, en ese momento cuando ustedes presionen la tecla “;” de su teclado en ingles, en la pantalla ven la tecla “ñ” pero si configuran en Windows el teclado en ingles veran la “;”, lo que quiero decir, es que en ambos caso el “scan code” es el mismo, lo que cambia es el codigo virtual.

Depues de esto, los programas pueden transforman dichos codigos en codigos del tipo Ascii y Unicode, por ejemplo en VB el evento “Keypress” muestra el codigo Ascii, pero el evento KeyDown y KeyUp muestran es el codigo virtual.

Ya teniendo el codigo virtual generado por Windows, el sistema operativo envia un mensaje WM_KEYDOWN, que en VB sería el evento “KeyDown”, a la ventana activa del hilo de la aplicación que se este ejecutando en ese momento. Al soltar dicha tecla se genera WM_KEYUP, que en VB sería “KeyDown”. Hay otros mensajes relacionado con el teclado, pero básicamente así es como funciona la interacción del teclado con la computadora.

El mouse, es un dispositivo mas sencillo, y no hay mucho que explicar, tan sencillamente hablamos de un puntero “hot spot”, el cual a medida que se mueve el mouse, dicho puntero se mueve, enviando mensajes constantemente al sistema operativo.

TECLADO Y MOUSE

Para este proyecto se necesita un formulario cuyo nombre sea “frmTeclado” y un modulo, dicho formulario tendra que contener los siguientes controles: 4 botones, 1 TextBox, 2 Label y un control Timer.

En formulario colocan este codigo:

Private Sub Command1_Click()

    Text1.SetFocus

    keybd_event VK_H, 0, 0, 0    ' press H

    keybd_event VK_H, 0, KEYEVENTF_KEYUP, 0   ' release H

    keybd_event VK_L, 0, 0, 0    ' press H

    keybd_event VK_L, 0, KEYEVENTF_KEYUP, 0   ' release H

End Sub

Private Sub Command2_Click()

    Label1.Caption = "PRESIONE ALT-A para desabilitar el evento del mouse"

    Timer1.Enabled = True

End Sub

Private Sub Command3_Click()

    Label1.Caption = "PRESIONE ALT-E para habilitar el puntero"

    ShowCursor False

End Sub

Private Sub Command4_Click()

    Label1.Caption = "PRESIONE ALT-H para liberar captura"

    SetCapture Me.hwnd

End Sub

Private Sub Form_KeyPress(KeyAscii As Integer)

    Me.Print Chr$(KeyAscii)

End Sub

Private Sub Form_Load()

    Randomize

    Timer1.Enabled = False

    Timer1.Interval = 500

    Command1.Caption = "Keyboard Event"

    Command1.Height = 510

    Command1.Left = 90

    Command1.Top = 45

    Command1.Width = 1770

    Command2.Caption = "Mouse Event"

    Command2.Height = 510

    Command2.Left = 90

    Command2.Top = 585

    Command2.Width = 1770

    Command3.Caption = "Desaparecer cursor"

    Command3.Height = 510

    Command3.Left = 90

    Command3.Top = 1125

    Command3.Width = 1770

    Command4.Caption = "Capturar Cursor"

    Command4.Height = 510

    Command4.Left = 90

    Command4.Top = 1665

    Command4.Width = 1770

    Label1.Caption = vbNullString

    Label1.Height = 690

    Label1.Left = 1935

    Label1.Top = 585

    Label1.Width = 1950

    Label2.Caption = vbNullString

    Label2.Height = 510

    Label2.Left = 1935

    Label2.Top = 1665

    Label2.Width = 1770

    Text1.Text = vbNullString

    Text1.Height = 510

    Text1.Left = 1935

    Text1.Top = 45

    Text1.Width = 1230

    Me.Width = 3990

    Me.Height = 2715

    RegisterHotKey Me.hwnd, ID_MOUSE, MOD_ALT, VK_A

    RegisterHotKey Me.hwnd, ID_CURSOR, MOD_ALT, VK_E

    RegisterHotKey Me.hwnd, ID_CAPTURE, MOD_ALT, VK_H

    nuevoGestor Me.hwnd

End Sub

Private Sub Form_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)

    Label2.Caption = "MOUSE SOLTADO EN " & X & " , " & Y

End Sub

Private Sub Form_Unload(Cancel As Integer)

    UnregisterHotKey Me.hwnd, ID_MOUSE

    UnregisterHotKey Me.hwnd, ID_CURSOR

    UnregisterHotKey Me.hwnd, ID_CAPTURE

End Sub

Private Sub Timer1_Timer()

    mouse_event MOUSEEVENTF_ABSOLUTE Or MOUSEEVENTF_MOVE, Int(Rnd() * Screen.Width * 15), Int(Rnd() * Screen.Height * 15), 0&, 0&

End Sub
En el modulo:

Public Const VK_A = 65

Public Const VK_H = 72

Public Const VK_E = 69

Public Const VK_L = 76

Declare Sub keybd_event Lib "user32.dll" (ByVal bVk As Byte, ByVal bScan As Byte, ByVal dwFlags As Long, ByVal dwExtraInfo As Long)

Declare Function RegisterHotKey Lib "user32" (ByVal hwnd As Long, ByVal id As Long, ByVal fsModifiers As Long, ByVal vk As Long) As Long

Declare Function UnregisterHotKey Lib "user32" (ByVal hwnd As Long, ByVal id As Long) As Long

Declare Function SetActiveWindow Lib "user32" (ByVal hwnd As Long) As Long

Declare Sub mouse_event Lib "user32" (ByVal dwFlags As Long, ByVal dx As Long, ByVal dy As Long, ByVal cButtons As Long, ByVal dwExtraInfo As Long)

Declare Function ShowCursor Lib "user32" (ByVal bShow As Long) As Long

Declare Function SetCapture Lib "user32" (ByVal hwnd As Long) As Long

Declare Function ReleaseCapture Lib "user32" () As Long

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Public Const GWL_WNDPROC = (-4)

Public Const WM_HOTKEY = &H312

Public Const MOD_ALT = &H1

Public Const MOD_CONTROL = &H2

Public Const MOD_SHIFT = &H4

Public Const MOD_WIN = &H8

Public Const ID_MOUSE = 1

Public Const ID_CURSOR = 2

Public Const ID_CAPTURE = 3

Dim ViejoGestor As Long

Public entro As Boolean

Public Const KEYEVENTF_KEYUP = &H2

Public Const MOUSEEVENTF_LEFTDOWN = &H2

Public Const MOUSEEVENTF_LEFTUP = &H4

Public Const MOUSEEVENTF_MIDDLEDOWN = &H20

Public Const MOUSEEVENTF_MIDDLEUP = &H40

Public Const MOUSEEVENTF_MOVE = &H1

Public Const MOUSEEVENTF_ABSOLUTE = &H8000

Public Const MOUSEEVENTF_RIGHTDOWN = &H8

Public Const MOUSEEVENTF_RIGHTUP = &H10

Public Sub nuevoGestor(hwnd As Long)

    ViejoGestor = SetWindowLong(hwnd, GWL_WNDPROC, AddressOf GestorMensaje)

End Sub

Public Function GestorMensaje(ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    Select Case mensaje

        Case WM_HOTKEY

            Select Case wParam

            Case ID_MOUSE

                frmteclado.Timer1.Enabled = False

                frmteclado.Label1.Caption = vbNullString

            Case ID_CURSOR

                frmteclado.Label1.Caption = vbNullString

                ShowCursor True

            Case ID_CAPTURE

                frmteclado.Label1.Caption = vbNullString

                frmteclado.Label2.Caption = vbNullString

                ReleaseCapture

            End Select

            SetActiveWindow hwnd

    End Select

    GestorMensaje = CallWindowProc(ViejoGestor, hwnd, mensaje, wParam, lParam)

End Function
Al correr el programa y presionar los botones veran cosas curiosas, con el boton “Keyboard event” veran como se escriben 2 letras en el textbox, sin presionar en el teclado. Con el boton “Mouse event”, notaran que el puntero del mouse se empieza a mover aleatoriamente en la pantalla, con el boton “Desaparecer cursor”, vemos como el puntero se esconde, y con el último botón capturamos el puntero del mouse en una ventana especifica.

Para entender todo esto, vamonos a las funciones.

keybd_event

Api: Declare Sub keybd_event Lib "user32.dll" (ByVal bVk As Byte, ByVal bScan As Byte, ByVal dwFlags As Long, ByVal dwExtraInfo As Long)

Esta función es la que se encarga de simular la pulsación de una tecla. Como primer parámetro pasamos el “codigo virtual” de dicha tecla, como segundo parámetro podemos pasar el “scan code”,  el parámetro “dwFlags”, puede tomar los siguientes valores:

Private Const KEYEVENTF_EXTENDEDKEY = &H1 

Private Const KEYEVENTF_KEYUP = &H2 : Indica que la tecla presionada se ha liberado.

La primera constante, no entiendo su uso, ya que es usada cuando se le pasa un valor al parámetro “bScan”, y en realidad en los ejemplos no uso dicho parámetro. Pero en la documentación oficial en MSDN encontraran el uso de dicha constante.

Como último parámetro se le pasa una información adicional de 32 Bits asociado a la tecla presionada.

En nuestro ejempo lo usamos de la siguiente manera:

keybd_event VK_H, 0, 0, 0    

keybd_event VK_H, 0, KEYEVENTF_KEYUP, 0   

En la primera linea simulamos “keyDown”, es decir, que se esta presionando, en estos caso pasamos el codigo virtual, y la segunda linea hace lo mismo pero coloco la constante KEYEVENTF_KEYUP para indicarle al sistema que la tecla ha dejado de ser presionada.

A continuación muestro la tabla de los codigos virtuales de la mayoria de las teclas, esta tabla fue extraida de la ayuda de MSDN de Microsoft.

	Symbolic constant name
	Value (hexadecimal)
	Mouse or keyboard equivalent

	VK_LBUTTON
	01
	Left mouse button 

	VK_RBUTTON
	02
	Right mouse button 

	VK_CANCEL
	03
	Control-break processing 

	VK_MBUTTON
	04
	Middle mouse button (three-button mouse) 

	— 
	05–07
	Undefined 

	VK_BACK
	08
	backspace key 

	VK_TAB
	09
	tab key 

	— 
	0A–0B
	Undefined 

	VK_CLEAR
	0C
	clear key 

	VK_RETURN
	0D
	enter key 

	— 
	0E–0F
	Undefined 

	VK_SHIFT
	10
	shift key 

	VK_CONTROL
	11
	ctrl key 

	VK_MENU
	12
	alt key 

	VK_PAUSE
	13
	pause key 

	VK_CAPITAL
	14
	caps lock key 

	— 
	15–19
	Reserved for Kanji systems 

	— 
	1A
	Undefined 

	VK_ESCAPE
	1B
	esc key 

	— 
	1C–1F
	Reserved for Kanji systems 

	VK_SPACE
	20
	spacebar 

	VK_PRIOR
	21
	page up key 

	VK_NEXT
	22
	page down key 

	VK_END
	23
	end key 

	VK_HOME
	24
	home key 

	VK_LEFT
	25
	left arrow key 

	VK_UP
	26
	up arrow key 

	VK_RIGHT
	27
	right arrow key 

	VK_DOWN
	28
	down arrow key 

	VK_SELECT
	29
	select key 

	— 
	2A
	Original equipment manufacturer (OEM) specific 

	VK_EXECUTE
	2B
	execute key 

	VK_SNAPSHOT
	2C
	print screen key 

	VK_INSERT
	2D
	ins key 

	VK_DELETE
	2E
	del key 

	VK_HELP
	2F
	help key 

	VK_0
	30
	0 key 

	VK_1
	31
	1 key 

	VK_2
	32
	2 key 

	VK_3
	33
	3 key 

	VK_4
	34
	4 key 

	VK_5
	35
	5 key 

	VK_6
	36
	6 key 

	VK_7
	37
	7 key 

	VK_8
	38
	8 key 

	VK_9
	39
	9 key 

	— 
	3A–40
	Undefined 

	VK_A
	41
	a key 

	VK_B
	42
	b key 

	VK_C
	43
	c key 

	VK_D
	44
	d key 

	VK_E
	45
	e key 

	VK_F
	46
	f key 

	VK_G
	47
	g key 

	VK_H
	48
	h key 

	VK_I
	49
	i key 

	VK_J
	4A
	j key 

	VK_K
	4B
	k key 

	VK_L
	4C
	l key 

	VK_M
	4D
	m key 

	VK_N
	4E
	n key 

	VK_O
	4F
	o key 

	VK_P
	50
	p key 

	VK_Q
	51
	q key 

	VK_R
	52
	r key 

	VK_S
	53
	s key 

	VK_T
	54
	t key 

	VK_U
	55
	u key 

	VK_V
	56
	v key 

	VK_W
	57
	w key 

	VK_X
	58
	x key 

	VK_Y
	59
	y key 

	VK_Z
	5A
	z key 

	VK_LWIN
	5B
	Left Windows key (Microsoft Natural Keyboard) 

	VK_RWIN
	5C
	Right Windows key (Microsoft Natural Keyboard) 

	VK_APPS
	5D
	Applications key (Microsoft Natural Keyboard) 

	— 
	5E–5F
	Undefined 

	VK_NUMPAD0
	60
	Numeric keypad 0 key 

	VK_NUMPAD1
	61
	Numeric keypad 1 key 

	VK_NUMPAD2
	62
	Numeric keypad 2 key 

	VK_NUMPAD3
	63
	Numeric keypad 3 key 

	VK_NUMPAD4
	64
	Numeric keypad 4 key 

	VK_NUMPAD5
	65
	Numeric keypad 5 key 

	VK_NUMPAD6
	66
	Numeric keypad 6 key 

	VK_NUMPAD7
	67
	Numeric keypad 7 key 

	VK_NUMPAD8
	68
	Numeric keypad 8 key 

	VK_NUMPAD9
	69
	Numeric keypad 9 key 

	VK_MULTIPLY
	6A
	Multiply key 

	VK_ADD
	6B
	Add key 

	VK_SEPARATOR
	6C
	Separator key 

	VK_SUBTRACT
	6D
	Subtract key 

	VK_DECIMAL
	6E
	Decimal key 

	VK_DIVIDE
	6F
	Divide key 

	VK_F1
	70
	f1 key 

	VK_F2
	71
	f2 key 

	VK_F3
	72
	f3 key 

	VK_F4
	73
	f4 key 

	VK_F5
	74
	f5 key 

	VK_F6
	75
	f6 key 

	VK_F7
	76
	f7 key 

	VK_F8
	77
	f8 key 

	VK_F9
	78
	f9 key 

	VK_F10
	79
	f10 key 

	VK_F11
	7A
	f11 key 

	VK_F12
	7B
	f12 key 

	VK_F13
	7C
	f13 key 

	VK_F14
	7D
	f14 key 

	VK_F15
	7E
	f15 key 

	VK_F16
	7F
	f16 key 

	VK_F17
	80H
	f17 key 

	VK_F18
	81H
	f18 key 

	VK_F19
	82H
	f19 key 

	VK_F20
	83H
	f20 key 

	VK_F21
	84H
	f21 key 

	VK_F22
	85H
	f22 key 

	VK_F23
	86H
	f23 key 

	VK_F24
	87H
	f24 key 

	— 
	88–8F
	Unassigned 

	VK_NUMLOCK
	90
	num lock key 

	VK_SCROLL
	91
	scroll lock key 

	— 
	92–B9
	Unassigned 

	— 
	BA–C0
	OEM specific 

	— 
	C1–DA
	Unassigned 

	— 
	DB–E4
	OEM specific 

	— 
	E5
	Unassigned 

	— 
	E6
	OEM specific 

	— 
	E7–E8
	Unassigned 

	— 
	E9–F5
	OEM specific 

	VK_ATTN
	F6
	Attn key

	VK_CRSEL
	F7
	CrSel key

	VK_EXSEL
	F8
	ExSel key

	VK_EREOF
	F9
	Erase EOF key

	VK_PLAY
	FA
	Play key

	VK_ZOOM
	FB
	Zoom key

	VK_NONAME
	FC
	Reserved for future use. 

	VK_PA1
	FD
	PA1 key

	VK_OEM_CLEAR
	FE
	Clear key


Ustedes podrian decir, que los valores usados en el proyecto son distinto a los de la tabla, lo que sucede es que en el proyecto use valores “decimales”, y estos valores se encuentran especificado en “hexadecimal”.

mouse_event

Api: Declare Sub mouse_event Lib "user32" (ByVal dwFlags As Long, ByVal dx As Long, ByVal dy As Long, ByVal cButtons As Long, ByVal dwExtraInfo As Long)

Esta función es la encargada de simular algun evento del mouse, como primer parametro “dwFlags” especificamos que acción vamos a realizar, para ello dicho parametro puede tomar estas constantes:

Public Const MOUSEEVENTF_LEFTDOWN = &H2 : Con este simulamos que el botón derecho del mouse fue presionado.

Public Const MOUSEEVENTF_LEFTUP = &H4 : Con este simulamos que el botón derecho del mouse fue dejado de presionar.

Public Const MOUSEEVENTF_MIDDLEDOWN = &H20 : Con este simulamos que el botón medio del mouse fue presionado.

Public Const MOUSEEVENTF_MIDDLEUP = &H40 : Con este simulamos que el botón medio del mouse fue dejado de presionar.

Public Const MOUSEEVENTF_MOVE = &H1 : Con esta constante indicamos que movemos el mouse.

Public Const MOUSEEVENTF_ABSOLUTE = &H8000 : Especificamos “el como” se realiza el movimiento del mouse.

Public Const MOUSEEVENTF_RIGHTDOWN = &H8 : Con este simulamos que el botón derecho del mouse fue presionado.

Public Const MOUSEEVENTF_RIGHTUP = &H10 : Con este simulamos que el botón derecho del mouse fue dejado de presionar.

Public Const MOUSEEVENTF_WHEEL = &H800 : Solo usado en NT, especifica que la rueda que poseen algunos mouse, ha sido movida, la cantidad del movimiento especificado en el parámetro “cButtons”.

Como segundo y tercer parámetro especificamos las coordenadas del puntero del mouse, en donde se realizara la acción especificada arriba.

El el cuarto parámetro especificamos la cantidad de movimiento de la rueda del mouse, si fue especificado en “dwFlags” la constante MOUSEEVENTF_WHEEL. Si el valor es positivo implica que la rueda fue movida hacia delante, si es negativo la rueda fue movida hacia atrás.

Y como último parámetro pasamos una información extra, asociada al evento del mouse.

Vamos a explicar la constante MOUSEEVENTF_ABSOLUTE, ya que en su explicación puse : (“el como” se realiza el movimiento del mouse”). Imaginemos que el puntero se encuentra en la coordenada (pixeles) 400x300, y llama la función mouse_event, con dx = 100 y dy = 50, si no especifico dicha constante, la nueva coordenada del mouse es 500x350, es decir, se desplaza, en cambio si uso dicha constante en la función, la nueva coordenada sería 100x50

En nuestro ejemplo usamos la constante, esa es la razón del por que el mouse cada 500 milisegundos cambia de posición:

mouse_event MOUSEEVENTF_ABSOLUTE Or MOUSEEVENTF_MOVE, Int(Rnd() * Screen.Width * 15), Int(Rnd() * Screen.Height * 15), 0&, 0&

Ustedes diran ¿Para que simular movimientos, clic, etc, del mouse, o simular las funciones del teclado?. Si ustedes desarrollaran por ejemplo un juego, sería perfecto, estas funciones, para hacer los “demos”, es decir, cuando ustedes estan jugando un juego, y en el menu del juego no hacen nada, algunas veces dicho juego, puede simular por ejemplo, un partido de futbol, dicha simulación puede estar simulando las teclas usadas para realizar dicha acción. Pero esto es un ejemplo, pueden haber muchos mas casos.

RegisterHotKey

Api: Declare Function RegisterHotKey Lib "user32" (ByVal hwnd As Long, ByVal id As Long, ByVal fsModifiers As Long, ByVal vk As Long) As Long

Hot key en español sería algo como “tecla caliente”, como su traduccion no es tan bonita me referire a ella como hot Key. Esta función lo que hace es registrar un hot key. ¿Qué es un hot key?. Cuando una ventana registra un hot key, dicha ventana recibe el mensaje, aun si la ventana no esta seleccionada, es decir, si yo registro como hot key la tecla “a”, mi ventana cada ves que se presiona la tecla “a” recibe dicho mensaje, sin importar si la tecla “a” fue presionada en otra aplicación, esto es muy importante a la hora de quererle dar prioridad a su aplicación, si se presionan cierta combinación de teclas.

Veamos los parámetros que tenemos: como primer parámetro le pasamos el “hwnd” de la ventana que recibira dicho mensajes, el segundo parámetro es un ID que identifica al mensaje, como tercer parámetro podemos pasarle las siguientes constantes:

Public Const MOD_ALT = &H1 : Implica que la tecla a recibir es ALT-X, donde X es la tecla especificada en el cuarto parametro “vk”.

Public Const MOD_CONTROL = &H2 : Implica que la tecla a recibir es CTRL-X, donde X es la tecla especificada en el cuarto parametro “vk”.

Public Const MOD_SHIFT = &H4 : Implica que la tecla a recibir es SHIFT-X, donde X es la tecla especificada en el cuarto parametro “vk”.

Y como último parámetro recibimos el codigo virtual, de la tecla o carácter que queremos registrar, veamos un ejemplo:

RegisterHotKey Me.hwnd, ID_CURSOR, MOD_ALT, VK_E

Aqui estoy registrando la combinacion de teclas ALT-E, es decir, no importa donde estes cuando presiones ALT-E, la ventana recibira dicho mensaje. Hay que dejar en claro que el parámetro ID tiene que ser unico para cada combinación.

¿Cómo recibimos el mensaje?. Estos mensajes requieren de un tratamiento especial, ya que cuando se presiona una combinación de teclas configuradas como “hot key”, Windows le envia a la ventana que lo registro el mensaje WM_HOTKEY. Que en nuestro modulo se encuentra especificado de esta manera:

Case WM_HOTKEY

            Select Case wParam            

            Case ID_MOUSE

                frmteclado.Timer1.Enabled = False

                frmteclado.Label1.Caption = vbNullString

            Case ID_CURSOR

                frmteclado.Label1.Caption = vbNullString

                ShowCursor True

            Case ID_CAPTURE

                frmteclado.Label1.Caption = vbNullString

                frmteclado.Label2.Caption = vbNullString

                ReleaseCapture

End Select

El parámetro “wParam” contiene el ID del mensaje que se activo, y el parámetro “lParam” se divide en dos, la parte baja “loword” contiene la información acerca del valor colocado en el parámetro “fsModifiers”, y la parte alta “hiword” contiene la tecla virtual o codigo virtual de la tecla presionada.

Ejemplo, corran el programa y presionen el botón “Desaparecer Cursor”, luego activen la ventana de VB u otra ventana, en este momento, la ventana de la aplicación desaparece, claro esta que todo es como incomodo por el puntero del mouse no esta, pero ahora presionen ALT-E, veran que el cursor aparece, y la ventana se activa, es decir, sin importar, que otra ventana estuviera activada, nuestra aplicación recibio el mensaje de la tecla hot key configurada. ¡impresionante!.

Claro esta, que la activacion de nuestra ventana no es gracia al hot key. Si no es gracia a la función localizada en el mensaje WM_HOTKEY, que se explica a continuación:

SetActiveWindow

Api: Declare Function SetActiveWindow Lib "user32" (ByVal hwnd As Long) As Long

Esta función se encarga de activar la ventana especificada en el “hwnd”.

UnregisterHotKey

Api Declare Function UnregisterHotKey Lib "user32" (ByVal hwnd As Long, ByVal id As Long) As Long

Esta función se encarga de eliminar la tecla registrada por la función RegisterHotKey, como primer parámetro le pasamos el “hwnd” de la ventana que la registro y como segundo parametro le pasamos el ID que registramos.

Pueden ver que la usamos en Form_Unload:

UnregisterHotKey Me.hwnd, ID_MOUSE

UnregisterHotKey Me.hwnd, ID_CURSOR

UnregisterHotKey Me.hwnd, ID_CAPTURE

ShowCursor

Api: Declare Function ShowCursor Lib "user32" (ByVal bShow As Long) As Long

Se encarga de mostrar u ocultar el cursor del mouse. Como unico parámetro se le pasa TRUE o FALSE, si es FALSE el cursor de esconde y en caso de ser TRUE el cursor se muestra.

SetCapture

Api: Declare Function SetCapture Lib "user32" (ByVal hwnd As Long) As Long

Esta función lo que hace es que el “hwnd” especificado como unico parametro recibe o captura todos los mensajes del mouse, pero esto no es lo que ustedes imaginan, es decir, esto no implica que si uso dicha función y doy click en otra aplciacion, mi ventana recibira dicho clic, esto no funciona exactamente así. Si trabajaramos con ventanas creadas en el mismo hilo de aplicación en teoria la ventana si recibiria dicho mensajes, pero cuando se le da clic a una ventana perteneciente a otro hilo de aplicación dicha captura se elimina.

¿Para que es importante esto?. Imaginemos que realizamos un programa que dibuja un cuadro, pero resulta que cuando creamos el cuadro (arrastrando el mouse por la pantalla), si nuestro mouse sale de la ventana al soltarlo en otro lugar que no sea nuestro ventana la pregunta sería ¿Quién captura el mensaje MouseUp?. Nuestra aplicación o la aplicación en donde soltamos el mouse, si tienes capturado el mouse dicho mensaje lo recibe nuestra ventana, en caso contrario no lo recibe.

Algunos que han trabajdo con esto podrían decir ¿Pero VB recibe el evento MouseUp, aun si el cursor no esta en nuestra ventana?. Eso quiere decir que el evento internamente sin que nosotros lo veamos esta capturando el mouse.

Pero hagan este ejemplo, corran el programa y presionen el botón “Capturar cursor” y den un clic afuera de la ventana y veran que es nuestra ventana quien recibe dicho mensaje, pero solo la recibe la primera vez, en caso de volver a presionar afuera de nuestra ventana nuestra aplicación pierde la captura del mouse. ¿Por qué funciona una vez?. Esta respusta no la se, en caso de que una persona le sepa le agradeceria dicha respuesta y con gusto la colocare en el manual.

Pero en teoria, o lo que dice la documentación oficial dice, es que dicha función captura los eventos del mouse en la ventana especificada como primer parámetro, tal vez, aunque no he realizado la prueba, dicho efecto, si vea mejor en apliacaciones que poseen varias ventanas hijas, pero esa prueba se las dejo a ustedes.

ReleaseCapture

Api: Declare Function ReleaseCapture Lib "user32" () As Long

Esta función libera la captura del mouse, en la ventana especificada en la función SetCapture,  este tambien se puede liberar si se activa otra ventana que pertenezca a otro hilo de aplicación.

OTRAS FUNCIONES SOBRE TECLADO Y MOUSE

SetFocus

Api: Declare Function SetFocus Lib "user32" (ByVal hwnd As Long) As Long

Función utilizada para darle el foco a una ventana que pertenezca al hilo de la aplicación, es decir, si tienen una ventana “A” que tiene controles,  con esta función, si la usan dentro de dicha ventana “A” pueden darle el foco a los controles que ella posee, si otra aplicación “B” tiene controles y ustedes saben los “hwnd” de dicho controles, si usan esta función en la ventana “A” con los “hwnd” de la ventana “B”, dicha acción no funciona.

Problema de esta función con VB. Resulta que en VB no se pueden usar esta función con este nombre. Ya que si en el formulario ponen:


SetFocus hwnd

El programa le genera un error, ya que da la casualidad que el formulario tiene una función con ese mismo nombre. Para ello usamos un alias, por ejemplo:

Declare Function PonerFoco Lib "user32" Alias "SetFocus" (ByVal hwnd As Long) As Long

Ahora si ponen en el formulario lo siguiente:


PonerFoco hwnd

Si! funciona, pueden ponerle cualquier nombre, siempre y cuando el “alias” posea el nombre original de la función.

GetFocus

Api: Declare Function GetFocus Lib "user32" () As Long

Este función retorna el “hwnd” de la ventana que actualmente tienen el foco.

GetActiveWindow

Api: Declare Function GetActiveWindow Lib "user32" () As Long

Función que retorna el “hwnd” de la ventana que esta activa.

BlockInput

Api: Declare Function BlockInput Lib "USER32.DLL" (ByVal fBlockIt As Long) As Long

Función que bloquea todos los mensajes de entrada, es decir, del mouse y teclado. Dichos mensajes son bloqueados cuando le pasamos a la función TRUE y para desbloquear y recibir los mensajes pasamos FALSE.

SwapMouseButton

Api: Declare Function SwapMouseButton Lib "user32" (ByVal bSwap As Long) As Long

Ustedes saben que windows, permite configurar los botones del mouse, que quiero decir, todos sabemos que el botón izquierdo, es el principal, es decir, es el botón con que nosotros activamos y hacemos las acciones principales, y el derecho es el botón secundario, ese orden puede ser invertido, y para ello usamos esta función.

Si le pasamos TRUE cambiamos dicho orden, y si pasamos FALSE restauramos el orden original.

SetDoubleClickTime

Api: Declare Function SetDoubleClickTime Lib "user32" (ByVal wCount As Long) As Long

Tambien sabemos que Windows permite configurar el “doble click”, es decir, con esta función configuramos el tiempo que debe de haber entre un click y otro click para activar el evento “Doble Clic”. Ese tiempo que lleva el parámetro “wCount” debe de estar en milisegundos.

GetDoubleClickTime

Api: Declare Function GetDoubleClickTime Lib "user32" () As Long

Esta función retorna la cantidad de milisegundos configurados actualemente para detectar el doble click.

CARET

Imaginen que un control del tipo TextBox o Caja de texto, tiene el foco, lo primero que ven es una rayita titilando, es decir,  como prendido y apagado, indicándonos en donde vamos a escribir, bueno esa rayita se llama “CARET”. Y tambien se han preguntado ¿Podre cambiar dicha rayita y poner otra cosa?. El proyecto que se muestra a continuación muestra como cambiar dicha rayita a cualquier cosa que nosotros queramos.

Para este proyecto se necesita, 1 botón, 1 caja de texto o TextBox, y un control timer. En el lugar donde corren el proyecto coloquen la imagen “caret.bmp” que acompaña a este manual. Y por último en el formulario copien este codigo:

Private Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal dwImageType As Long, ByVal dwDesiredWidth As Long, ByVal dwDesiredHeight As Long, ByVal dwFlags As Long) As Long

Private Declare Function ShowCaret Lib "user32" (ByVal hwnd As Long) As Long

Private Declare Function HideCaret Lib "user32" (ByVal hwnd As Long) As Long

Private Declare Function CreateCaret Lib "user32" (ByVal hwnd As Long, ByVal hBitmap As Long, ByVal nWidth As Long, ByVal nHeight As Long) As Long

Private Const IMAGE_BITMAP = 0

Private Const LR_LOADFROMFILE = &H10

Dim bitmap As Long

Private Sub Command1_Click()

    Timer1.Enabled = False

End Sub

Private Sub Form_Load()

    Command1.Caption = "Mostrar Caret"

    Command1.Left = 90

    Command1.Top = 405

    Command1.Height = 510

    Command1.Width = 2355

    Text1.Text = vbNullString

    Text1.Left = 90

    Text1.Top = 45

    Text1.Height = 285

    Text1.Width = 2355

    Timer1.Enabled = True

    Timer1.Interval = 5000

    Me.Width = 2610

    Me.Height = 1395

End Sub

Private Sub Text1_GotFocus()

    bitmap = LoadImage(App.hInstance, App.Path & "\caret.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

    r = CreateCaret(Text1.hwnd, bitmap, 0, 0)

    ShowCaret Text1.hwnd

End Sub

Private Sub Timer1_Timer()

    HideCaret Text1.hwnd

End Sub
Al arrancar dicha aplicación, denle el foco al textbox y se daran cuenta que tiene un “caret” diferente, a los 5 segundos el timer se activa y lo desaparece, para que se vuelva activar deben de presionar el botón y luego volver a darle el foco al control. Claro esta, como pueden ver no me esforcé en hacer un bonito “caret” pero estoy seguro que su creatividad sera mucho mejor que la mia.

CreateCaret

Api: Declare Function CreateCaret Lib "user32" (ByVal hwnd As Long, ByVal hBitmap As Long, ByVal nWidth As Long, ByVal nHeight As Long) As Long

Esta función se utiliza para  crear un nuevo caret y asignarselo a una ventana. Como primer parámetro tenemos a dicha ventana, como segundo parámetro le pasamos el bitmap que queremos mostrar, si dicho parámetro le pasamos NULL o 0, se establece un caret solido, en caso de ser 1 el caret es gris. En caso de especificar una imagen, los dos ultimos parámetros son ignorados. En caso contrario, los dos ultimos parámetros especifican el alto y ancho del caret, estas unidades son logicas.

En nuestro ejemplo:

bitmap = LoadImage(App.hInstance, App.Path & "\caret.bmp", IMAGE_BITMAP, ByVal 0&, ByVal 0&, LR_LOADFROMFILE)

 r = CreateCaret(Text1.hwnd, bitmap, 0, 0)

ShowCaret Text1.hwnd

En la primera linea, cargo la imagen, luego se la asigno a la ventana y por último muestro el caret.

ShowCaret

Api: Declare Function ShowCaret Lib "user32" (ByVal hwnd As Long) As Long

Función encargada de mostrar el caret de la ventana especificada en su primer parametro.

HideCaret

Api: Declare Function HideCaret Lib "user32" (ByVal hwnd As Long) As Long

Función encargada de ocultar el caret de la ventana especificada en su primer parametro.

MULTIMEDIA

Windows posee de una cantidad muy importante de funciones API’S para menejar los diferentes tipos de multimedias, como video, VCR, midi, wav, cd-audio, etc. Particularmente no voy a enfocar dicha seccion a explicar esas API’S, increiblemente Windows nos proporciona  2 funciones con las cuales podemos hacer praticamente TODO!!. Digo “practicamente” ya que tampoco abarca todo. Pero en mi opinion, podrian crear muy facil un programa parecido a Winamp con tan solo 1 función.

Esas dos funciones reciben el nombre de mciSendCommand y mciSendString. Estas funciones hacen lo mismo, pero de diferente maneras ¿Qué significa esto?. La primera trabaja en base a puro comandos (constantes) y la segunda trabaja mediante cadenas de texto.

Seguramente estan confundido como una función puede realizar tanto trabajo y ciertamente es sorprendente, pero todo lo que voy a explicar requiere de su concentración, por que tan poco es tan facilito entender dicho funcionamiento.

¿Qué significa mensajes en “cadena de texto”?. Esta función mciSendString es la que mas me gusta, ya que es muy parecido al lenguaje humano. Es decir, para hacer funcionar o darle play a un archivo .wav es necesario algo por el estilo.

Cadena = “play xxxxx.wav from 0 to final”. E español sería algo como: “Ejecuta xxxx.wav desde el milisegundo 0 hasta ‘final’ (final = que es un valor) .

En cambio mciSendCommand, funciona como funcionaria un robot, es decir, le das valores y la función hace algo. Cuando entremos de lleno, con dichas explicaciónes verán la diferencia. Así que empecemos con la primera, que es la mas sencilla, ya que si logran entender dicha función les sera mas facíl entender mciSendCommand.
MCISENDSTRING
mciSendString
Api: Declare Function mciSendString Lib "winmm.dll" Alias "mciSendStringA" (ByVal lpstrCommand As String, ByVal lpstrReturnString As String, ByVal uReturnLength As Long, ByVal hwndCallback As Long) As Long

Como primer parametro “lpstrCommand”, tenemos la cadena de texto, que especifica una acción, como segundo parametro “lpstrReturnString”, tenemos una cadena de texto que nos sirve cuando dicha acción, nos retorna un valor a tomar en cuenta, como tercer parametro “uReturnLength” tenemos la cantidad de caracteres que queremos agarrar en el parametro anterior, y como último parametro tenemos “hwndCallBack” aquí (ojo: no es obligatorio) podemos especificar un “hwnd” de una ventana cualquiera, esta ventana cuando es especificada y la acción tiene cierto parametro, el gestor de mensajes de dicha ventana recibira un mensaje especial, de todas manera eso lo veremos mas adelante.

La pregunta de ustedes sería ¿Cómo formo la cadena para que haga algo util?. A partir de ahora voy a empezar a explicar palabras claves que necesitarán para formar su “oración” para así ejecutar lo que ustedes quieran.

Antes que nada, hay que entender cierto “protocolo”, es decir, los pasos básicos a realizar para ejecutar alguna acción sobre un archivo. Windows puede tener cierto dispositivos “device” configurados, esos dispositivos pueden ser:

cdaudio = utilizado para escuchar la musica de los CD.

digitalvideo  = video digital.
overlay = video analogo.

scanner = scanear imágenes.

sequencer = MIDI

vcr = para trabajar con video caseteras

videodisc = para trabajar con aparatos de video, en donde insertas discos compactos.

waveaudio = trabajar con archivo .WAV

Nosotros en este apartado solo trabajaremos con los dispositivos “waveaudio”, “cdaudio” y “digitalvideo”.

Lo que mencione arriba es que antes de trabajar con un dispositivo, hay que abrirlo, para ello es el mensaje:

OPEN

¿Cómo es la sintaxis? Sería algo así:

"open %s1 %s2 %s3"

“%s” indica que ahi va algo, “ese algo” se explica continuación:

%s1 = en nuestro caso aquí va el nombre del archivo. Que vamos abrir.

%s2 = aquí van los llamados “flags”, ¿Qué es eso?. Los flags son como indicadores, que le informan a la función, de algunos detalles. Estos “flag” su uso dependera del dispositivo que este abierto, es decir, hay “flag” que solo se usan en dispositivos del tipo “waveaudio” que no sirven en “vcr”.

A continuación menciono los tipos de flag para cada dispositivo:

	Cdaudio
	alias device_alias
shareable
	type device_type

	Digitalvideo
	alias device_alias
elementname
nostatic
parent hwnd
shareable
	style child
style overlapped
style popup
style style_type
type device_type

	Overlay
	alias device_alias
parent hwnd
shareable
style child
	style overlapped
style popup
style style_type
type device_type

	Sequencer
	alias device_alias
shareable
	type device_type

	Vcr
	alias device_alias
shareable
	type device_type

	Videodisc
	alias device_alias
shareable
	type device_type

	waveaudio
	alias device_alias
buffer buffer_size
	shareable
type device_type


Mostrando dicha tabla, que fue tomada de la ayuda de MSDN de microsoft, aclaro, que no me voy a concentrar en explicar todos los “flag” explicare aquellos que considere mas importante, de todas maneras en la ayuda de MSDN encontraran la información completa.

alias device_alias = especifica algún alias, que queramos usar.

type device_type = aquí se específica el dispositivo el cual queremos cargar.

shareable = Inicializa, el dispositivo o archivo de manera compartida, es decir, mas de un dispositivo puede abrirse al mismo tiempo.

style child = abre una ventana con el estilo CHILD

style overlapped = abre una ventana con el estilo OVERLAPPED

style popup = abre la ventana con el estilo POPUP.

%s3 = aquí podemos colocar las palabras “wait” , “notify”, o ambos, la explicación de que hacen dichas palabras se explicaran mas adelante.

Ya visto para que sirve cada “sintaxis” hagamos un ejemplo, imaginen que queremos abrir el archivo “xxxx.wav”.

El mensaje o la cadena de mensaje sería algo así:

“open xxxx.wav type waveaudio alias archivowav wait”


Como pueden ver, se pueden activar mas de un “flag” a la vez, en nuestro caso usamos 2: type  y  alias.

Cuando ya tenemos abierto un dispositivo, estamos listo para manipularlos, a continuación veremos algunos comandos o mensajes que permiten manipular dicho dispositivo abierto.

CAPABILITY

Sintaxis: "capability %s1 %s2 %s3"

Con este comando podemos obtener información, o la capacidad de un dispositivo para realizar una acción.

%s1  = Aquí colocamos al identificador del dispositivo, si usan “alias”, iria el nombre de dicho alias.

%s2  = pueden ir los siguiente indicadores o “flag”:

	cdaudio
	can eject
can play
can record
can save
compound device
	device type
has audio
has video
uses files

	digitalvideo
	can eject
can freeze
can lock
can play
can record
can reverse
can save
can stretch
can stretch input
can test
	compound device
device type
has audio
has still
has video
maximum play rate
minimum play rate
uses files
uses palettes
windows

	overlay
	can eject
can freeze
can play
can record
can save
can stretch
	compound device
device type
has audio
has video
uses files
windows

	sequencer
	can eject
can play
can record
can save
compound device
	device type
has audio
has video
uses files

	vcr
	can detect length
can eject
can freeze
can monitor sources
can play
can preroll
can preview
can record
can reverse
can save
can test
	clock increment rate
compound device
device type
has audio
has clock
has timecode
has video
number of marks
seek accuracy
uses files

	videodisc
	can eject
can play
can record
can reverse
can save
CAV
CLV
compound device
	device type
fast play rate
has audio
has video
normal play rate
slow play rate
uses files

	waveaudio
	can eject
can play
can record
can save
compound device
device type
	has audio
has video
inputs
outputs
uses files


La mayoria de estos indicadores, retornan TRUE o FALSE. Por ejemplo:

“capability nombrealias can eject”

Si fuera “nombrealias” un dispositivo cdaudio , este comando nos retornaría TRUE o FALSE indicándonos que la unidad de CD-ROM puede ejecutar “eject”, es decir, sacar la bandeja donde va el CD.

¿Dónde se retorna dicho valor?. Se recuerdan del parámetro de la función mciSendString llamado “lpstrReturnString”, pues aquí es donde se almacena dicho valor.

En particular creo que los flag, se explican por si solos, así que me evito entrar en detalles.

%s3 = aquí podemos colocar las palabras “wait” , “notify”, o ambos
STATUS

Sintaxis: “status %s1 %s2 %s3"

Nos sirve para retornar información acerca del estado de un dispositivo, y son muchos tipos de estado y lo veremos en las flag que usa dicho comando.

%s1  = Aquí colocamos al identificador del dispositivo, si usan “alias”, iria el nombre de dicho alias.

%s2  = pueden ir los siguiente indicadores o “flag”:

	cdaudio
	cdaudio type track number
current track
length
length track number
media present
mode
	number of tracks
position
position track number
ready
start position
time format

	digitalvideo
	audio
audio alignment
audio bitspersample
audio breaks
audio bytespersec
audio input
audio record
audio source
audio samplespersec
audio stream
bass
bitsperpel
brightness
color
contrast
current track
disk space drive
file completion
file format 
file mode
forward
frames skipped
gamma
input
left volume
length
length track number
media present
mode
monitor
monitor method
nominal
nominal frame rate
nominal record frame rate
number of tracks
	output
palette handle
pause mode
play speed
position
position track number
ready
record frame rate
reference frame
reserved size
right volume
seek exactly
sharpness
smpte
speed
start position
still file format
time format
tint
treble
unsaved
video
video key index
video key color
video record
video source
video source number
video stream
volume
window handle
window visible
window minimized
window maximized

	overlay
	media present
mode
number of tracks
	ready
stretch
window handle

	sequencer
	current track
division type
length
length track number
master
media present
mode
number of tracks
offset
	port
position
position track number
ready
slave
start position
tempo
time format

	vcr
	assemble record
audio monitor
audio monitor number
audio record 
audio record track number
audio source
audio source number
channel
channel tuner number
clock
clock id
counter
counter format
counter resolution
current track
frame rate
index
index on
length
length track number
media present
media type
mode
number of audio tracks
number of tracks
number of video tracks
	pause timeout
play format
position
position start
position track number
postroll duration
power on
preroll duration
ready
record format
speed
time format
time mode
time type
timecode present
timecode record
timecode type
tuner number
video monitor
video monitor number
video record
video record track number
video source
video source number
write protected

	videodisc
	current track
disc size
forward
length
length track number
media present
media type
mode
	number of tracks
position
position track number
ready
side
speed
start position
time format

	waveaudio
	alignment
bitspersample
bytespersec
channels
current track
format tag
input
length
length track number
level
	media present
mode
number of tracks
output
position
position track number
ready
samplespersec
start position
time format


Como pueden ver son muchos así que explicare los que considero mas importante:

audio = retorna TRUE o FALSE, donde TRUE significa que las cornetas estan activadas.

audio bytespersec = retorna el average de bytes por segundos usados para grabar.

audio samplepersec = retorna el número de trazos o “muestra” por segundos.

bass = retorna el actual nivel del audio-bass.

channel = retorna un valor entero del canal activo.

cannels = retorna 1 si los canales estan en mono y 2 si estan en stereo.

left volume = retorna el volumen del canal izquierdo.

right volume = retorna el volumen del canal derecho.

lenght =  retorna la longitud de un archivo media, esta longitud se retorna en el formato actual de tiempo, que quiero decir, un archivo .wav retornaría la cantidad de milisegundos que posee dicha canción o sonido. El formato en que lo regresa depende del formato de tiempo configurado, eso lo veremos cuando veamos el comando SET.

position = retorna la posición actual. Esta posición depende del formato del tiempo. Colocando el mismo ejemplo del archivo .wav, este flag nos retornaría en que posición, en ese tiempo, se esta escuchando.

power on  = retorna TRUE si el VCR esta encendido.

ready = retorna TRUE si el dispositivo esta listo para recibir otro comando.

speed  = retorna la velocidad del dispositivo.

time format = retorna el tipo de format del tiempo.

volume = retorna un promedio o average del volumen izuiqerdo y derecho.

%s3 = aquí podemos colocar las palabras “wait” , “notify”, o ambos.
La información que retorna dicho comando es colocado en la variable “lpstrReturnString” que se le pasa a la función.

CLOSE

Sintaxis: “close %s1 %s2"

Este comando nos sirve para cerrar un dispositivo que este abierto.

%s1  = Aquí colocamos al identificador del dispositivo, si usan “alias”, iria el nombre de dicho alias.

%s2 = aquí podemos colocar las palabras “wait” , “notify”, o ambos.

Ejm: “close nombrealias”

PLAY

Sintaxis: “play %s1 %s2 %s3”

Este comando ejecuta el “play” de un dispositivo.

%s1  = Aquí colocamos al identificador del dispositivo, si usan “alias”, iria el nombre de dicho alias.

%s2  = pueden ir los siguiente indicadores o “flag”:

	cdaudio
	from position 
	to position

	digitalvideo
	from position 
fullscreen
repeat
	Reverse
to position
window

	sequencer
	from position 
	to position

	vcr
	at time
from position 
reverse
	scan
to position

	videodisc
	fast
from position 
reverse
scan
	slow
speed integer
to position

	waveaudio
	from position 
	to position


from position = se especifica la posicion inicial donde se empieza a ejecutar el dispositovo. Por ejemplo, si un archivo .wav posee 5000 milisegundos de sonido, y hacemos esto “play nombrealias from 2500”, estamos empezando a escuchar el sonido en la mitad.

to position = indica la posición final hasta donde se ejecuta dicho dispositivo.

speed integer = con esto configuramos la velocidad en que se reproduce un videodisc.

fast = se especifica que la ejecución se debe hacer lo mas rapido posible, esta velocidad viene determinado por el flag “speed”.

fullscreen = especifica que la ejecución debería de hacerse a pantalla completa.

repeat = este flag indica que cuando se finalice la ejecución del dispositivo, este vuelve a empezar desde el comienzo.

reverse = la ejecución se hace en sentido contrario.

%s3 = aquí podemos colocar las palabras “wait” , “notify”, o ambos.
Ejm: “play nombrealias from 0”

PAUSE

Sintaxis: “pause %s1 %s2"

Este comando nos sirve para poner en pausa un dispositivo que se este ejecutando.

%s1  = Aquí colocamos al identificador del dispositivo, si usan “alias”, iria el nombre de dicho alias.

%s2 = aquí podemos colocar las palabras “wait” , “notify”, o ambos.

Ejm: “pause nombrealias”

RESUME

Sintaxis: “resume %s1 %s2"

Este comando nos sirve para continuar un dispositivo que este en pausa.

%s1  = Aquí colocamos al identificador del dispositivo, si usan “alias”, iria el nombre de dicho alias.

%s2 = aquí podemos colocar las palabras “wait” , “notify”, o ambos.

Ejm: “resume nombrealias”

SET

Sintaxis: “set %s1 %s2 %s3"

Este comando nos sirve para configurar o cambiar la configuración de un dispositivo.
%s1  = Aquí colocamos al identificador del dispositivo, si usan “alias”, iria el nombre de dicho alias.

%s2  = pueden ir los siguiente indicadores o “flag”:

	cdaudio
	audio all off
audio all on
audio left off
audio left on
audio right off
audio right on
	door closed
door open
time format milliseconds
time format msf
time format tmsf

	digitalvideo
	audio all off
audio all on
audio left off
audio left on
audio right off
audio right on
door closed
door open
	file format format
seek exactly on
seek exactly off
speed factor
still file format format
time format frames
time format milliseconds
video off
video on

	overlay
	audio all off
audio all on
audio left off
audio left on
audio right off
	audio right on
door closed
door open
video off
video on

	sequencer
	audio all off
audio all on
audio left off
audio left on
audio right off
audio right on
door closed
door open
master MIDI
master none
master SMPTE
offset time
	port mapper
port none
port port_number
slave file
slave MIDI
slave none
slave SMPTE
tempo tempo_value
time format milliseconds
time format SMPTE fps
time format SMPTE 30 drop
time format song pointer

	vcr
	assemble record on
assemble record off
audio all off
audio all on
audio left off
audio left on
audio right off
audio right on
clock time
counter format
counter value
door closed
door open
index counter
index date
index time
index timecode
length duration
pause timeout
postroll duration -
duration
	power on
power off
preroll duration duration
record format SP
record format LP
record format EP
speed factor
time format frames
time format hms
time format milliseconds
time format msf
time format SMPTE fps
time format SMPTE 30 drop
time format tmsf
time mode counter
time mode detect
time mode timecode
tracking plus
tracking minus
tracking reset

	videodisc
	audio all off
audio all on
audio left off
audio left on
audio right off
audio right on
door closed
	door open
time format frames
time format hms
time format milliseconds
time format track
video off
video on

	waveaudio
	alignment integer
any input
any output
audio all off
audio all on
audio left off
audio left on
audio right off
audio right on
bitspersample bit_count
bytespersec byte_rate
	channels channel_count
door closed
door open
format tag pcm
format tag tag
input integer
output integer
samplespersec integer
time format bytes
time format milliseconds
time format samples


audio all on,  audio all off = estos flag se encarga de desactivar o activar los cornetas o la salida de audio.

audio left on,  audio left off, audio right on,  audio right off: se encargan de desactivar o activar la salida de sonido, de cada canal ya sea el canal derecho o izquierdo.

door open = abre o expulsa, la bandeja donde se coloca el CD

door close = cierra, la bandeja donde se coloca el CD

input integer = especifica la cantidad canales de salida a utilizar.

power on, power off  = prende o apaga el dispositivo.

time format bytes  = Asigna el tiempo en formato de bytes, por ejemplo, si trabajamos con una archivo .wav, dicho formato retornaría el tamaño en bytes del archivo.

time format frames =  Asigna el tiempo en formato de “frames” o marcos, por ejemplo una película se puede especificar en este formato y saber por cual marco o frame se esta ejecutando.

time format hms =  Asigna el formato del tiempo en HH:MM:SS, donde HH = horas, MM = minutos y SS = segundos.

time format milliseconds =  Asigna el formato del tiempo en milisegundos.

time format mfs =  Asigna el formato del tiempo en MM:SS:FF, donde MM = minutos, SS  = segundos y FF = frames o marcos.

video on , video off =  activa o desactiva la salida de video.

%s3 = aquí podemos colocar las palabras “wait” , “notify”, o ambos.
Ejm: “set nombrealias door open”, en caso de que “nombrealias” sea un dispositivo cdaudio este abriría la puerta del CD-ROM.

STOP

Sintaxis: “stop %s1 %s2 %s3"

Este comando nos sirve para detener la ejecución de un dispositivo.

%s1  = Aquí colocamos al identificador del dispositivo, si usan “alias”, iria el nombre de dicho alias.

%s2 = aquí se utiliza un solo flag, que es “hold” dicho flag solo funciona para digitalvideo.
%s3 = aquí podemos colocar las palabras “wait” , “notify”, o ambos.

Ejm: “stop nombrealias”

Estos son los comando básicos mas importantes a conocer, puede que a medida que salen diferentes versiones de Windows  estos comandos y los flags de los comandos ya existentes se amplíen, por eso recomiendo revisar la documentación oficial de MSDN en internet para estar al actualizado. La explicación de dichos comando y flags fueron básicos, ya que eso es lo que voy a enseñar sobre la función mciSendString. Mas adelante veremos un ejemplo en VB que trabaja con dicho mensajes.

Debi mencionar que en la documentación de MSDN tambien podran encontrar mensajes que solo sirven para ciertos dispositivos, es decir, hay mensajes solamente para digitalvideo. Estos mensajes se les llama “mensajes extendidos” (Extended Command).

WAIT Y NOTIFY

En todos los comando abran visto esta línea: “aquí podemos colocar las palabras “wait” , “notify”, o ambos.”
¿Qué significa poner wait?. Si ponemos este mensaje:

“play dispositivo from 0 to 100 wait”, esto implica que el comando “play” ¡no! retornara el control a la aplicación hasta que se ejecute dicho “play”.

¿Qué significa poner notify?. Si ponemos este mensaje:

“play dispositivo from 0 to 100 notify”, esto implica que al finalizar dicha ejecución o “play”, se le envia a la ventana de la aplicación el mensaje MM_MCINOTIFY, ¿Qué ventana recibe el mensaje?. Es la ventana que se especifica en el parámetro dwCallBack de la función. Cabe mencionar que para recibir dicho mensaje en VB tendría que usar Subclassing. Este mensaje les sirve para saber si cada comando se ejecuta de manera correcta. 

ERRORES

¿Qué sucede si hay un error, como lo capturamos?. Con el mensaje MM_MCINOTIFY podemos saber si la ejecución fue correcta aunque no es necesario capturar simpre dicho mensaje, pero como saber ¿Cuál es el error?. Para ello existe una una función llamada:

mciGetErrorString

Api: Declare Function mciGetErrorString Lib "winmm.dll" Alias "mciGetErrorStringA" (ByVal dwError As Long, ByVal lpstrBuffer As String, ByVal uLength As Long) As Long

Con esta función retornamos una cadena de texto que identifica a dicho error. Como primer parámetro le pasamos el número de error, como segundo parámetro le pasamos la variable de tipo String que almacenara el texto que describe al error, y como tercer parámetro le pasamos la cantidad de espacio que queremos recuperar.

¿Cómo saber el número de error?. Cuando ejecutan la función mciSendString el valor que ella retorna es el “número de error”. Ejm:

resultado = mciSendString(...,...,.....)

Esta variable “resultado” es la que se coloca en el parámetro “dwError” de la función.

SONIDO WAV (MCISENDSTRING)
Ya visto la teoria, vamos al proyecto que usa dicha función. Primero dicho proyecto tendra que contener los siguientes componentes: Microsoft Common Dialog Control 6.0, Microsoft Windows Common Controls 6.0 (sp4). Con estos coponentes activados, procedemos a incluir los controles que presento a continuación:

4 Botones (Command).

2 Etiquetas (Label).

1 ProgressBar.

1 Slider 

1 Caja de Texto (TextBox)

1 Timer

Luego pegan este código en el formulario:

Private Declare Function mciSendString Lib "winmm.dll" Alias "mciSendStringA" (ByVal lpstrCommand As String, ByVal lpstrReturnString As String, ByVal uReturnLength As Long, ByVal hwndCallback As Long) As Long

Private Declare Function mciGetErrorString Lib "winmm.dll" Alias "mciGetErrorStringA" (ByVal dwError As Long, ByVal lpstrBuffer As String, ByVal uLength As Long) As Long

Dim sendstring As String

Dim resultado As Long

Dim error As String

Dim cadena As String

Dim slider As Boolean

Private Sub Command1_Click()

    With CommonDialog1

        .Filter = "Archivos *.wav | *.wav"

        .ShowOpen

        Text1.Text = .FileName

    End With

    MandarMensaje "close archivowav"

    MandarMensaje "open " & Text1.Text & " type waveaudio alias archivowav"

    MandarMensaje "set archivowav time format milliseconds"

    MandarMensaje "status archivowav length"

    If Val(Trim(cadena)) <> 0 Then

        ProgressBar1.Min = 1

        ProgressBar1.Max = Val(cadena)

        ProgressBar1.Value = 1

        Slider1.Min = 1

        Slider1.Max = Val(cadena)

        Slider1.Value = 1

        Label2.Caption = formatotiempo(Val(cadena))

    End If

End Sub

Private Sub Command2_Click()

    Timer1.Enabled = True

    MandarMensaje "play archivowav from 0"

End Sub

Sub BuscarError(número As Long)

    error = Space(255)

    mciGetErrorString resultado, error, 255

    Label1.Caption = error

End Sub

Private Sub Command3_Click()

    MandarMensaje "stop archivowav"

    Timer1.Enabled = False

End Sub

Private Sub Command4_Click()

    If Command4.Caption = "Pausa" Then

        MandarMensaje "pause archivowav"

        Command4.Caption = "Continuar"

    Else

        MandarMensaje "resume archivowav"

        Command4.Caption = "Pausa"

    End If

End Sub

Sub MandarMensaje(mensaje As String)

        cadena = Space(255)

        resultado = mciSendString(mensaje, cadena, Len(cadena), 0)

        BuscarError resultado

End Sub

Private Sub Slider1_Change()

    If Slider1.Value = Slider1.Max Then

        Debug.Print "ENTRO"

        Timer1.Enabled = False

    End If

End Sub

Private Sub Slider1_MouseDown(Button As Integer, Shift As Integer, x As Single, y As Single)

    slider = True

End Sub

Private Sub Slider1_MouseUp(Button As Integer, Shift As Integer, x As Single, y As Single)

    slider = False

    MandarMensaje "play archivowav from " & Str(Slider1.Value) & " to " & Str(Slider1.Max)

End Sub

Private Sub Timer1_Timer()

    MandarMensaje "status archivowav position"

    ProgressBar1.Value = Val(cadena)

    If Not slider Then

        Slider1.Value = Val(cadena)

    End If

    Label2.Caption = formatotiempo(ProgressBar1.Max - Val(cadena))

End Sub

Function formatotiempo(milisegundos As Long) As String

    Dim segundos As Integer

    Dim minutos As Integer

    segundos = milisegundos \ 1000

    minutos = segundos \ 60

    segundos = segundos Mod 60

    formatotiempo = Str(minutos) & ":" & Str(segundos)

End Function

Private Sub Form_Load()

    Command1.Caption = "Archivo WAV"

    Command1.Height = 375

    Command1.Left = 3825

    Command1.Top = 180

    Command1.Width = 1185

    Command2.Caption = "Play"

    Command2.Height = 510

    Command2.Left = 90

    Command2.Top = 675

    Command2.Width = 1185

    Command3.Caption = "Stop"

    Command3.Height = 510

    Command3.Left = 1305

    Command3.Top = 675

    Command3.Width = 1185

    Command4.Caption = "Pausa"

    Command4.Height = 510

    Command4.Left = 2520

    Command4.Top = 675

    Command4.Width = 1185

    Label1.Caption = ""

    Label1.Height = 1140

    Label1.Left = 225

    Label1.Top = 2160

    Label1.Width = 7170

    Label2.Caption = ""

    Label2.Height = 195

    Label2.Left = 7155

    Label2.Top = 1890

    Label2.Width = 1185

    ProgressBar1.Height = 195

    ProgressBar1.Left = 270

    ProgressBar1.Top = 1665

    ProgressBar1.Width = 8250

    Slider1.Height = 360

    Slider1.Left = 90

    Slider1.Top = 1305

    Slider1.Width = 8520

    Text1.Height = 330

    Text1.Left = 90

    Text1.Top = 180

    Text1.Width = 3570

    Text1.Text = vbNullString

    Timer1.Enabled = False

    Timer1.Interval = 100

    Me.Width = 8880

    Me.Height = 3495

End Sub

Cuando arranquen el proyecto, aprietan el botón “Archivo WAV” y buscan en su computadora un archivo *.wav, después que lo seleccionen proceden a escucharlo, con los botones “PLAY”, “STOP” y “PAUSE”. ¿Cómo funciona todo esto?. El codigo lo voy explicar desde que se carga el archivo .wav, pero primero expliquemos 2 funciones claves:

Sub MandarMensaje(mensaje As String)

        cadena = Space(255)

        resultado = mciSendString(mensaje, cadena, Len(cadena), 0)

        BuscarError resultado

End Sub

Dicha función envia un mensaje con la función mciSendString, si el mensaje retorna un resultado, dicha información se almacena en “cadena”, y el codigo del error en “resultado”, pueden ver que en el cuarto parámetro no paso ningun “hwnd”, por lo que es facil deducir que no uso el flag “notify”. La variable “resultado” es pasada a otra función “BuscarError”, que se encarga de mostrar el mensaje de error.

Sub BuscarError(número As Long)

    error = Space(255)

    mciGetErrorString resultado, error, 255

    Label1.Caption = error

End Sub
Pueden ver que capturo un máximo de 255 caracteres, y dicha “cadena” de error se almacena en la variable “error” que luego se le asigna al Caption del Label1. 

Ahora vamos a explicar pasa a paso lo que hacemos para que dicho programa funcione. Para ello empecemos cuando buscamos el archivo .wav en el evento Command1_Click():

Private Sub Command1_Click()


.


.


.

    MandarMensaje "close archivowav"


Cerramos cualquier dispositivo abierto que tenga el nombre del alias a usar.

    MandarMensaje "open " & Text1.Text & " type waveaudio alias archivowav"


Abrimos el archivo .wav que se especifica en el Text del control Text1, pueden ver que especificamos el type y el alias en los flag.

    MandarMensaje "set archivowav time format milliseconds"


Configuro el tiempo del archivo en formato de milisegundos.

    MandarMensaje "status archivowav length"


Busco cuanto tiempo, posee dicho archivo en milisegundos.

    If Val(Trim(cadena)) <> 0 Then ‘ Si dicho tiempo es distinto de 0

        ProgressBar1.Min = 1

        ProgressBar1.Max = Val(cadena)

        ProgressBar1.Value = 1

        Slider1.Min = 1

        Slider1.Max = Val(cadena)

        Slider1.Value = 1


Configuro los controles.

        Label2.Caption = formatotiempo(Val(cadena))


Muestro en formato “MM:SS” el tiempo en milisegundo que posee la variable “cadena”

    End If

End Sub
Si presionan el botón “play” ejecutamos lo siguiente:

Private Sub Command2_Click()

    Timer1.Enabled = True


Activo el timer, el cual permite ir avanzando en los controles Progress1 y Slider1

    MandarMensaje "play archivowav from 0"


Ejecuto “play” del archivo cuyo alias es “archivowav” desde (from) la posición 0.

End Sub

Nos vamos ahora al Timer:

Private Sub Timer1_Timer()

    MandarMensaje "status archivowav position"


Solicito la información sobre la posicion actual que se esta ejecutando, es decir, por cual milisegundo esta actualmente.

    ProgressBar1.Value = Val(cadena)

    If Not slider Then

        Slider1.Value = Val(cadena)

    End If

    
Le paso dicho valores a los controles, es por esa razon que se ve la barra moviendose.

    Label2.Caption = formatotiempo(ProgressBar1.Max - Val(cadena))


Actualizo el tiempo que falta.

End Sub

Si presionan “Stop” se ejecuta lo siguiente:

Private Sub Command3_Click()

    MandarMensaje "stop archivowav"


Paro la ejecución del archivo 

    Timer1.Enabled = False


Desactivo el Timer

End Sub
Si presionan “Pausa” se ejecuta lo siguiente:

Private Sub Command4_Click()

    If Command4.Caption = "Pausa" Then


Si el Caption es “Pausa” entoces

        MandarMensaje "pause archivowav"


Mando el mensaje “pause” para parar la ejecución del archivo.

        Command4.Caption = "Continuar"


Cambio el Caption del botón.

    Else


Si es “continuar”

        MandarMensaje "resume archivowav"


Mando el mensaje “resume” indicando que continue la ejecución del archivo.

        Command4.Caption = "Pausa"


Y coloco ”Pausa” en el Caption.

    End If

End Sub
¿Difícil?, no creo, pueden ver que al usar mciSendString el trabajo se hace muy natural, ya que los mensajes son intuitivos al momento de crearlos. Ahora vamos ver la función mciSendCommand y veremos el mismo proyecto con dicha función, y verán que la cosa se complica un poco.

MCISENDCOMMAND
mciSendCommand
Api: Declare Function mciSendCommand Lib "winmm.dll" Alias "mciSendCommandA" (ByVal wDeviceID As Long, ByVal uMessage As Long, ByVal dwParam1 As Long, ByRef dwParam2 As Any) As Long

Se podria decir que esta función es como mas “tecnica”, es decir, no tiene esa facilidad que posee mciSendString que consiste en usar una cadena de texto, en este función dicha cadena no existe, lo que existen son mensajes en forma de constante, y cuando queremos recuperar o asignar algun valor hay que hacerlo por medio de estructuras. Es decir, en pocas palabras esta función es mas complicada que la anterior.

Dichas constantes no las voy a explicar todas al igual que las estructuras, inclusive en esta función mi explicación va ser mas basica, y me voy concentrar en que entiendan el como funciona la función y no para que es cada variable, ya que si saben como trabaja esta función, ya las constante es cuestion de leer un poquito y listo!.

Como primer parámetro “wDeviceID”, se le pasa un valor que identifica a nuestro dispositivo, solamente cuando usamos el mensaje MCI_OPEN, en este parámetro ponemos 0, para cualquier otro mensaje dicho parámetro debe de tener un valor. Es casi lo equivalente al “alias” usado en mciSendString. Como segundo parámetro “uMessage”, le pasamos el mensaje que queremos ejecutar, aquí los mensajes no son cadenas de caracteres sino constantes, a continuación muestro un listado de los mensajes:

	Mensaje en mciSendString
	Mensaje en mciSendCommand

	break
	MCI_BREAK

	sysinfo
	MCI_SYSINFO

	capability
	MCI_GETDEVCAPS

	close
	MCI_CLOSE

	info
	MCI_INFO

	open
	MCI_OPEN

	status
	MCI_STATUS

	load
	MCI_LOAD

	pause
	MCI_PAUSE

	play
	MCI_PLAY

	record
	MCI_RECORD

	resume
	MCI_RESUME

	save
	MCI_SAVE

	seek
	MCI_SEEK

	set
	MCI_SET

	stop
	MCI_STOP


Pueden ver que puse los mensajes con sus respectivo “string”, es decir, mientras en mciSendString usamos el comando o mensaje “play” su equivalente en mciSendCommand es la constante o mensaje MCI_PLAY. A continuación muestro los valores de dichas constantes:

Private Const MCI_BREAK = &H811

Private Const MCI_SYSINFO = &H810

Private Const MCI_GETDEVCAPS = &H80B

Private Const MCI_CLOSE = &H804

Private Const MCI_INFO = &H80A

Private Const MCI_OPEN = &H803

Private Const MCI_STATUS = &H814

Private Const MCI_LOAD = &H850

Private Const MCI_PAUSE = &H809

Private Const MCI_PLAY = &H806

Private Const MCI_RECORD = &H80F

Private Const MCI_RESUME = &H855

Private Const MCI_SAVE = &H813

Private Const MCI_SEEK = &H807

Private Const MCI_SET = &H80D

Private Const MCI_STOP = &H808

Como tercer parametro “dwParam1” tenemos los flag o indicadores que acompañan al mensaje. Y como cuarto parámetro tenemos “dwParam2”, que es donde pasamos las estructuras que usamos ya sea para configurar o recuperar información.

A simple vista pueden ver que es muy diferente a la función vista con anterioridad. Ahora vamos a ver como funciona cada mensaje. Voy a trabajar los mismos mensajes o comandos estudiados en mciSendString.
MCI_OPEN

Cuando usamos este mensaje los parámetros de la función mciSendCommand puede tomar los siguientes valores:

wDeviceID = Particularmente cuando yo uso este mensaje pongo 0, ya que no hay dispositivo activado en ese momento.

UMessage = MCI_OPEN.

wParam1 = el uso de las flag puede ser un poco enredado pero presten atención, algunos de los valores que puede tomar son:

Private Const MCI_OPEN_ALIAS = &H400&

Private Const MCI_OPEN_SHAREABLE = &H100&

Private Const MCI_OPEN_TYPE = &H2000&

Private Const MCI_OPEN_TYPE_ID = &H1000&

Private Const MCI_OPEN_ELEMENT = &H200&

Private Const MCI_OPEN_ELEMENT_ID = &H800&

Flags exclusivos para el dispositivo digitalvideo.

Private Const MCI_DGV_OPEN_16BIT = &H80000&

Private Const MCI_DGV_OPEN_32BIT = &H100000&

Private Const MCI_DGV_OPEN_NOSTATIC = &H40000&

Private Const MCI_DGV_OPEN_PARENT = &H20000&

Private Const MCI_DGV_OPEN_WS = &H10000&

Flags exclusivos para el dispositivo overlay.

Private Const MCI_OVLY_OPEN_PARENT = &H20000

Private Const MCI_OVLY_OPEN_WS = &H10000

Flags exclusivos para el dispositivo waveaudio.

Private Const MCI_WAVE_OPEN_BUFFER = &H10000

Los complicados de los flag, es ¿Como usarlos?. Pero antes de explicar como se usan vamos a ver que toma el otro parámetro.

WParam2 = dicho parámetro toma una estructura del tipo MCI_OPEN_PARMS o una estructura de un dispositivo en especifico. Pero nosostros, en este mensaje trabajamos con MCI_OPEN_PARMS que posee la siguiente configuración:

Private Type MCI_OPEN_PARMS


dwCallback As Long ‘ Equivalente al hwdnCallback de mciSendString.


wDeviceID As Long ‘ ID del dispositivo que en nuestro caso al inicio sera 0.

lpstrDeviceType As String ‘ Nombre del dispositivo.


lpstrElementName As String ‘ Nombre del elemento, usualmente es la dirección del archivo que se quiere abrir


lpstrAlias As String ‘ Alias del dispositivo abierto.
End Type

¿Como usamos todos esto? Eso depende si entienden el uso de los flag ¿Qué quiero decir? Algunos flag indican un estado pero otros indican que se esta pasando información a traves de la estructura del cuarto parámetro. Por ejemplo

mciSendCommand(0, MCI_OPEN, MCI_OPEN_SHAREABLE or MCI_OPEN_ALIAS, openestructura).
Suponiendo que “openestructura” es una variable del tipo MCI_OPEN_PARMS, lo que significa la linea de arriba es que la función espera que “lpstrAlias” tenga información ¿Por qué espera eso? La función espera dicha información por que se activa el flag “MCI_OPEN_ALIAS “que  debe de tomar en cuenta dicha variable de la estructura. Ejemplo mas claro:

Openestructura.lpstrAlias = “nombre”

mciSendCommand(0, MCI_OPEN, MCI_OPEN_SHAREABLE, openestructura). 

En este ejemplo la función no toma en cuenta el valor que posee “lpstrAlias” y el por que, se debe a que no fue indicado el flag para que tomara en cuenta dicha información, volviendo a poner la siguiente linea:

Openestructura.lpstrAlias = “nombre”

mciSendCommand(0, MCI_OPEN, MCI_OPEN_SHAREABLE or MCI_OPEN_ALIAS, openestructura). 

Arreglamos el problema. Otro ejemplo:

Openestructura.lpstrAlias = “nombre”

Openestructura.lpstrElementName = direccionarchivo

mciSendCommand(0, MCI_OPEN, MCI_OPEN_SHAREABLE or MCI_OPEN_ALIAS, openestructura). 

De esta manera la función no toma el valor de “lpstrElementName” ya que no se ha especificado lo contrario, por lo que el dispositivo seguramente retornaría un error, el problema se soluciona de esta manera:

Openestructura.lpstrAlias = “nombre”

Openestructura.lpstrElementName = direccionarchivo

mciSendCommand(0, MCI_OPEN, MCI_OPEN_SHAREABLE or MCI_OPEN_ALIAS or MCI_OPEN_ELEMENT, openestructura). 

La pregunta inmediata que deben de tener es ¿El flag MCI_OPEN_SHAREABLE tiene algo que ver con la estructura? La respuesta es no!, estos son flag indicativos o que se usan para configurar algo. Dicho flag hace lo mismo que el flag “shareable” de “open” de la función mciSendString.

Segunda pregunta inmediata ¿Cómo saber que flag sirven para indicar algo nada mas, y cuales flag le indican a la función que debe de tomar en cuenta un valor de la estructura? Y es aquí donde tienen que saberse documentar, ya que en la ayuda de MSDN se especifica el uso de cada flag, y se darán cuenta de los flag que se usan para que la función tome en cuenta los valores de la estructura, a continuación muestro los flag que se necesitan para que la función tome en cuenta dicho valores:

Private Type MCI_OPEN_PARMS


dwCallback As Long ‘ Este valor es tomado en cuenta sin necesidad de algun flag.

wDeviceID As Long ‘Este valor es tomado en cuenta sin necesidad de algun flag.

lpstrDeviceType As String ‘ MCI_OPEN_TYPE o MCI_OPEN_TYPE_ID


lpstrElementName As String ‘ MCI_OPEN_ELEMENT o MCI_OPEN_ELEMENT_ID


lpstrAlias As String ‘ MCI_OPEN_ALIAS
End Type

Cuando se termina de ejecutar el mensaje MCI_OPEN, la variables “wDeviceID” de la estructura contiene el ID que vamos usar en todos los demas comandos que se encargan de manipular dicho dispositivo que fue abierto.

¿Donde ponemos el WAIT o NOTIFY? Ambos vienen especificados con las siguientes constantes:

Private Const MCI_NOTIFY = &H1&

Private Const MCI_WAIT = &H2&

Y se colocan en “wParam1”. Cabe mencionar que el mensaje “notify” sera enviado a la gestor de mensajes de la ventana especificada en “dwCallBack” .

Dichas constantes en TODOS los mensajes o comandos van en el parámetro “wParam1”.

CAPABILITY

Cuando usamos este mensaje los parámetros de la función mciSendCommand puede tomar los siguientes valores:

wDeviceID = Aquí colocamos “wDeviceID” de la estructura MCI_OPEN_PARMS, después de ejecutar el comando MCI_OPEN

uMessage = MCI_GETDEVCAPS.

wParam1 = Voy a poner los flag, que sirven para cualquier dispositivo, les dejo a ustedes documentarse sobre los otros flag:

Private Const MCI_GETDEVCAPS_DEVICE_TYPE = &H4&

Private Const MCI_GETDEVCAPS_HAS_AUDIO = &H2&

Private Const MCI_GETDEVCAPS_HAS_VIDEO = &H3&

Private Const MCI_GETDEVCAPS_ITEM = &H100&

Private Const MCI_GETDEVCAPS_CAN_EJECT = &H7&

Private Const MCI_GETDEVCAPS_CAN_PLAY = &H8&

Private Const MCI_GETDEVCAPS_CAN_RECORD = &H1&

Private Const MCI_GETDEVCAPS_CAN_SAVE = &H9&

WParam2 = Pasamos la estructura:

Private Type MCI_GETDEVCAPS_PARMS


dwCallback As Long ‘ Este parametro hace lo mismo que  el de la estructura del mensaje MCI_OPEN


dwReturn As Long


dwItem As Long

End Type

¿Como usamos los flag con la estructura? Sabemos que este mensaje MCI_GETDEVCAPS sirve para recuperar información, es intutivo pensar que la variable “dwReturn”, es quien contendra el resultado de dicha información, cuando explique el mensaje “capability” arriba, dije que los flag que acompañaban a dicho mensaje retornaban TRUE o FALSE, pues aquí sigue siendo igual, “dwReturn” en la mayoría de los casos retorna TRUE o FALSE, menciono mayoría ya que hay pequeñas excepciones.  Aquí no hay flag, que funcionan como en MCI_OPEN, es decir, no hay flag que indiquen a la función que tome en cuenta el valor de la estructura. Es facil pensar que esto se debe a que este comando sirve para recibir información.

Pero hay un hecho curioso con el uso de “dwIitem”, veamos un ejemplo:

capestructura.dwItem = MCI_GETDEVCAPS_CAN_EJECT
mciSendCommand(openestructura.wDeviceID , MCI_GETDEVCAPS, MCI_GETDEVCAPS_ITEM, capestructura). 

Ustedes ven, que arriba hay unas constante en azul, esto significa que cuando la estructura en su parámetro “wParam1” posee MCI_GETDEVCAPS_ITEM esto implica que en el parámetro “dwItem” de la estructura va una de las constante que estan azul. Esto no es valido:

mciSendCommand(openestructura.wDeviceID, MCI_GETDEVCAPS, MCI_GETDEVCAPS_ITEM  or MCI_GETDEVCAPS_CAN_EJECT, capestructura). 

Otro ejemplo:

mciSendCommand(openestructura.wDeviceID , MCI_GETDEVCAPS, MCI_GETDEVCAPS_DEVICE_TYPE , capestructura). 

Pueden ver que a la estructura “capestructura” no se le configura nada, el valor que retorna se almacena en “dwReturn” y no es ni FALSE ni TRUE, estos son los valores que puede retornar:

Private Const MCI_ALL_DEVICE_ID = - 1

Private Const MCI_DEVTYPE_ANIMATION = 519

Private Const MCI_DEVTYPE_CD_AUDIO = 516

Private Const MCI_DEVTYPE_DAT = 517

Private Const MCI_DEVTYPE_DIGITAL_VIDEO = 520

Private Const MCI_DEVTYPE_OTHER = 521

Private Const MCI_DEVTYPE_OVERLAY = 515

Private Const MCI_DEVTYPE_SCANNER = 518

Private Const MCI_DEVTYPE_SEQUENCER = 523

Private Const MCI_DEVTYPE_VCR = 513

Private Const MCI_DEVTYPE_VIDEODISC = 514

Private Const MCI_DEVTYPE_WAVEFORM_AUDIO = 522

Es decir, usamos este flag para saber que tipo de dispositivo es “openestructura.wDeviceID”
STATUS

wDeviceID = Aquí colocamos “wDeviceID” de la estructura MCI_OPEN_PARMS, después de ejecutar el comando MCI_OPEN

uMessage = MCI_STATUS.

wParam1 = Voy a poner los flag, que sirven para cualquier dispositivo, les dejo a ustedes documentarse sobre los otros flag:

Private Const MCI_STATUS_ITEM = &H100&

Private Const MCI_STATUS_CURRENT_TRACK = &H8&

Private Const MCI_STATUS_LENGTH = &H1&

Private Const MCI_STATUS_MODE = &H4&

Private Const MCI_STATUS_NUMBER_OF_TRACKS = &H3&

Private Const MCI_STATUS_POSITION = &H2&

Private Const MCI_STATUS_READY = &H7&

Private Const MCI_STATUS_TIME_FORMAT = &H6&

Private Const MCI_STATUS_START = &H200&

wParam2 = Pasamos la estructura:

Private Type MCI_STATUS_PARMS


dwCallback As Long


dwReturn As Long


dwItem As Long


dwTrack As Integer

End Type

Las constantes en azul funcionan exactamente igual que en el comando anterior. Y el valor que retorna “dwReturn” dependera de que mensaje se envie por ejemplo:

Si usamos la constante MCI_STATUS_MODE, “dwReturn” puede retornar los siguientes valores:

Private Const MCI_STRING_OFFSET = 512

Private Const MCI_MODE_NOT_READY = (MCI_STRING_OFFSET + 12)

Private Const MCI_MODE_OPEN = (MCI_STRING_OFFSET + 18)

Private Const MCI_MODE_PAUSE = (MCI_STRING_OFFSET + 17)

Private Const MCI_MODE_PLAY = (MCI_STRING_OFFSET + 14)

Private Const MCI_MODE_RECORD = (MCI_STRING_OFFSET + 15)

Private Const MCI_MODE_SEEK = (MCI_STRING_OFFSET + 16)

Private Const MCI_MODE_STOP = (MCI_STRING_OFFSET + 13)

Si usamos la constante MCI_STATUS_TIME_FORMAT, “dwReturn” puede retornar los siguientes valores:

Private Const MCI_FORMAT_BYTES = 8

Private Const MCI_FORMAT_FRAMES = 3

Private Const MCI_FORMAT_HMS = 1

Private Const MCI_FORMAT_MILLISECONDS = 0

Private Const MCI_FORMAT_MSF = 2

Private Const MCI_FORMAT_SAMPLES = 9

Private Const MCI_FORMAT_TMSF = 10

CLOSE

wDeviceID = Aquí colocamos “wDeviceID” de la estructura MCI_OPEN_PARMS, después de ejecutar el comando MCI_OPEN

uMessage = MCI_CLOSE.

wParam1 = Cuando usamos MCI_CLOSE, lo unico que podemos poner aquí son las constante del mensaje “notify” o “wait”.

wParam2 = en este parámetro pasamos la estructura:

Private Type MCI_GENERIC_PARMS


dwCallback As Long

End Type

Pueden ver que este mensaje es sencillo, esta estructura MCI_GENERIC_PARMS se usa en los mensajes o comandos sencillo, y lo único que se pasa es el HWDN de la ventana que capturara el mensaje “notify” en caso que se haya especificado dicha constante en “wParam1”.

PLAY

wDeviceID = Aquí colocamos “wDeviceID” de la estructura MCI_OPEN_PARMS, después de ejecutar el comando MCI_OPEN

uMessage = MCI_PLAY.

wParam1 = Voy a poner los flag, que sirven para cualquier dispositivo, les dejo a ustedes documentarse sobre los otros flag:

Private Const MCI_FROM = &H4&

Private Const MCI_TO = &H8&

wParam2 = en este parámetro pasamos la estructura:

Private Type MCI_PLAY_PARMS


dwCallback As Long


dwFrom As Long


dwTo As Long

End Type

Es intituitivo pensar, con los conocimientos ya visto, que se queremos que tome en cuenta los valores “dwFrom” y “dwTo”, tendremos que especificar en el parámetro “wParam1” los constantes vistas arriba.

PAUSE

wDeviceID = Aquí colocamos “wDeviceID” de la estructura MCI_OPEN_PARMS, después de ejecutar el comando MCI_OPEN

uMessage = MCI_PAUSE.

wParam1 = Cuando usamos MCI_CLOSE, lo unico que podemos poner aquí son las constante del mensaje “notify” o “wait”.

wParam2 = en este parámetro pasamos la estructura:

Private Type MCI_GENERIC_PARMS


dwCallback As Long

End Type

RESUME

wDeviceID = Aquí colocamos “wDeviceID” de la estructura MCI_OPEN_PARMS, después de ejecutar el comando MCI_OPEN

uMessage = MCI_RESUME.

wParam1 = Cuando usamos MCI_CLOSE, lo unico que podemos poner aquí son las constante del mensaje “notify” o “wait”.

wParam2 = en este parámetro pasamos la estructura:

Private Type MCI_GENERIC_PARMS


dwCallback As Long

End Type

STOP

wDeviceID = Aquí colocamos “wDeviceID” de la estructura MCI_OPEN_PARMS, después de ejecutar el comando MCI_OPEN

uMessage = MCI_STOP.

wParam1 = Cuando usamos MCI_CLOSE, lo único que podemos poner aquí son las constante del mensaje “notify” o “wait”.

wParam2 = en este parámetro pasamos la estructura:

Private Type MCI_GENERIC_PARMS


dwCallback As Long

End Type

SET

wDeviceID = Aquí colocamos “wDeviceID” de la estructura MCI_OPEN_PARMS, después de ejecutar el comando MCI_OPEN

uMessage = MCI_SET.

wParam1 = Voy a poner los flag, que sirven para cualquier dispositivo, les dejo a ustedes documentarse sobre los otros flag:

Private Const MCI_SET_AUDIO = &H800&

Private Const MCI_SET_AUDIO_ALL = &H4001&

Private Const MCI_SET_AUDIO_LEFT = &H4002&

Private Const MCI_SET_AUDIO_RIGHT = &H4003&

Private Const MCI_SET_DOOR_CLOSED = &H200&

Private Const MCI_SET_DOOR_OPEN = &H100&

Private Const MCI_SET_OFF = &H4000&

Private Const MCI_SET_ON = &H2000&

Private Const MCI_SET_TIME_FORMAT = &H400&

Private Const MCI_FORMAT_BYTES = 8

Private Const MCI_FORMAT_FRAMES = 3

Private Const MCI_FORMAT_HMS = 1

Private Const MCI_FORMAT_MILLISECONDS = 0

Private Const MCI_FORMAT_MSF = 2

Private Const MCI_FORMAT_SAMPLES = 9

Private Const MCI_FORMAT_TMSF = 10

Private Const MCI_SET_VIDEO = &H1000&

wParam2 = Pasamos la estructura:

Private Type MCI_SET_PARMS


dwCallback As Long


dwTimeFormat As Long


dwAudio As Long

End Type

Vamos a explicar mediante ejemplo como funciona las constantes marcadas en azul:

setestructura.dwAudio = MCI_SET_AUDIO_ALL 
mciSendCommand(openestructura.wDeviceID , MCI_SET, MCI_SET_AUDIO or MCI_SET_ON, setestructura)

En el caso del tiempo, es parecido:

setestructura.dwTimeFormat = MCI_FORMAT_MILLISECONDS 

mciSendCommand(openestructura.wDeviceID , MCI_SET, Private Const MCI_SET_TIME_FORMAT = &H400&

 , setestructura)

Con esto cubri los mensajes todos los mensajes que explique en mciSendString, se daran cuenta que en estos mensajes o constantes, no me preocupe por explicar que hacia cada constante, mi enfoque era que entendieran en como usar dichas constante ya que eso es lo mas difícil de entender, algunas constante cumplen la misma definición que los flag que usa mciSendString.

Al igual que mciSendString, la función mciSendCommand lo que retorna es el valor que se usa en la función mciGetErrorString.

SONIDO WAV (MCISENDCOMMAND)
Ya visto la teoria, vamos al proyecto que usa dicha función. Primero dicho proyecto tendra que contener los siguientes componentes: Microsoft Common Dialog Control 6.0, Microsoft Windows Common Controls 6.0 (sp4). Con esto coponentes activados, procedemos a incluir los controles que presento a continuación:

4 Botones (Command).

2 Etiquetas (Label).

1 ProgressBar.

1 Slider 

1 Caja de Texto (TextBox)

1 Timer

Luego pegan este codigo en el formulario:

Private Declare Function mciSendCommand Lib "winmm.dll" Alias "mciSendCommandA" (ByVal wDeviceID As Long, ByVal uMessage As Long, ByVal dwParam1 As Long, ByRef dwParam2 As Any) As Long

Private Declare Function mciGetErrorString Lib "winmm.dll" Alias "mciGetErrorStringA" (ByVal dwError As Long, ByVal lpstrBuffer As String, ByVal uLength As Long) As Long

Private Const MCI_OPEN = &H803

Private Const MCI_OPEN_ALIAS = &H400&

Private Const MCI_OPEN_ELEMENT = &H200&

Private Const MCI_OPEN_TYPE = &H2000&

Private Const MCI_SET = &H80D

Private Const MCI_FORMAT_MILLISECONDS = 0

Private Const MCI_SET_TIME_FORMAT = &H400&

Private Const MCI_STATUS_LENGTH = &H1&

Private Const MCI_STATUS_ITEM = &H100&

Private Const MCI_STATUS = &H814

Private Const MCI_STATUS_POSITION = &H2&

Private Const MCI_FROM = &H4&

Private Const MCI_TO = &H8&

Private Const MCI_STOP = &H808

Private Const MCI_PAUSE = &H809

Private Const MCI_RESUME = &H855

Private Const MCI_PLAY = &H806

Private Const MCI_CLOSE = &H804

Private Type MCI_PLAY_PARMS

    dwCallback As Long

    dwFrom As Long

    dwTo As Long

End Type

Private Type MCI_OPEN_PARMS

    dwCallback As Long

    wDeviceID As Long

    lpstrDeviceType As String

    lpstrElementName As String

    lpstrAlias As String

End Type

Private Type MCI_STATUS_PARMS

    dwCallback As Long

    dwReturn As Long

    dwItem As Long

    dwTrack As Integer

End Type

Private Type MCI_SET_PARMS

    dwCallback As Long

    dwTimeFormat As Long

    dwAudio As Long

End Type

Dim sendstring As String

Dim resultado As Long

Dim error As String

Dim cadena As String

Dim slider As Boolean

Dim mciopen As MCI_OPEN_PARMS

Dim mcistatus As MCI_STATUS_PARMS

Dim mciplay As MCI_PLAY_PARMS

Dim mciset As MCI_SET_PARMS

Private Sub Command1_Click()

    With CommonDialog1

        .Filter = "Archivos *.wav | *.wav"

        .ShowOpen

        Text1.Text = .FileName

    End With

    mciopen.wDeviceID = 0

    mciopen.lpstrAlias = "archivowav1"

    mciopen.lpstrDeviceType = "waveaudio"

    mciopen.lpstrElementName = Text1.Text

    resultado = mciSendCommand(0, MCI_OPEN, MCI_OPEN_ALIAS Or MCI_OPEN_TYPE Or MCI_OPEN_ELEMENT, mciopen)

    BuscarError resultado

    mciset.dwTimeFormat = MCI_FORMAT_MILLISECONDS

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_SET, MCI_SET_TIME_FORMAT, 0)

    BuscarError resultado

    mcistatus.dwItem = MCI_STATUS_LENGTH

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_STATUS, MCI_STATUS_ITEM, mcistatus)

    BuscarError resultado

    If mcistatus.dwReturn <> 0 Then

        ProgressBar1.Min = 1

        ProgressBar1.Max = mcistatus.dwReturn

        ProgressBar1.Value = 1

        Slider1.Min = 1

        Slider1.Max = mcistatus.dwReturn

        Slider1.Value = 1

        Label2.Caption = formatotiempo(mcistatus.dwReturn)

    End If

End Sub

Private Sub Command2_Click()

    Timer1.Enabled = True

    mciplay.dwFrom = 0

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_PLAY, MCI_FROM, mciplay)

    BuscarError resultado

End Sub

Sub BuscarError(número As Long)

    error = Space(255)

    mciGetErrorString resultado, error, 255

    Label1.Caption = error

End Sub

Private Sub Command3_Click()

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_STOP, 0, 0)

    BuscarError resultado

    Timer1.Enabled = False

End Sub

Private Sub Command4_Click()

    If Command4.Caption = "Pausa" Then

        resultado = mciSendCommand(mciopen.wDeviceID, MCI_PAUSE, 0, 0)

        BuscarError resultado

        Command4.Caption = "Continuar"

    Else

        resultado = mciSendCommand(mciopen.wDeviceID, MCI_RESUME, 0, 0)

        BuscarError resultado

        Command4.Caption = "Pausa"

    End If

End Sub

Private Sub Form_Load()

        Command1.Caption = "Archivo WAV"

    Command1.Height = 375

    Command1.Left = 3825

    Command1.Top = 180

    Command1.Width = 1185

    Command2.Caption = "Play"

    Command2.Height = 510

    Command2.Left = 90

    Command2.Top = 675

    Command2.Width = 1185

    Command3.Caption = "Stop"

    Command3.Height = 510

    Command3.Left = 1305

    Command3.Top = 675

    Command3.Width = 1185

    Command4.Caption = "Pausa"

    Command4.Height = 510

    Command4.Left = 2520

    Command4.Top = 675

    Command4.Width = 1185

    Label1.Caption = ""

    Label1.Height = 1140

    Label1.Left = 225

    Label1.Top = 2160

    Label1.Width = 7170

    Label2.Caption = ""

    Label2.Height = 195

    Label2.Left = 7155

    Label2.Top = 1890

    Label2.Width = 1185

    ProgressBar1.Height = 195

    ProgressBar1.Left = 270

    ProgressBar1.Top = 1665

    ProgressBar1.Width = 8250

    Slider1.Height = 360

    Slider1.Left = 90

    Slider1.Top = 1305

    Slider1.Width = 8520

    Text1.Height = 330

    Text1.Left = 90

    Text1.Top = 180

    Text1.Width = 3570

    Text1.Text = vbNullString

    Timer1.Enabled = False

    Timer1.Interval = 100

    Me.Width = 8880

    Me.Height = 3495

End Sub

Private Sub Form_Unload(Cancel As Integer)

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_CLOSE, 0, ByVal 0&)

    BuscarError resultado

End Sub

Private Sub Slider1_Change()

    If Slider1.Value = Slider1.Max Then

        Timer1.Enabled = False

    End If

End Sub

Private Sub Slider1_MouseDown(Button As Integer, Shift As Integer, x As Single, y As Single)

    slider = True

End Sub

Private Sub Slider1_MouseUp(Button As Integer, Shift As Integer, x As Single, y As Single)

    slider = False

    mciplay.dwFrom = Slider1.Value

    mciplay.dwTo = Slider1.Max

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_PLAY, MCI_FROM Or MCI_TO, mciplay)

    BuscarError resultado

End Sub

Private Sub Timer1_Timer()

    mcistatus.dwItem = MCI_STATUS_POSITION

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_STATUS, MCI_STATUS_ITEM, mcistatus)

    BuscarError resultado

    ProgressBar1.Value = mcistatus.dwReturn

    If Not slider Then

        Slider1.Value = mcistatus.dwReturn

    End If

    Label2.Caption = formatotiempo(ProgressBar1.Max - mcistatus.dwReturn)

End Sub

Function formatotiempo(milisegundos As Long) As String

    Dim segundos As Integer

    Dim minutos As Integer

    segundos = milisegundos \ 1000

    minutos = segundos \ 60

    segundos = segundos Mod 60

    formatotiempo = Str(minutos) & ":" & Str(segundos)

End Function

Pueden ver que hace exactamente lo mismo, con la diferencia que use mciSendCommand, ahora vamos a ir explicando pasa a paso lo que se hizo:

Private Sub Command1_Click()

   
.


.

.


    mciopen.wDeviceID = 0

    mciopen.lpstrAlias = "archivowav1"

    mciopen.lpstrDeviceType = "waveaudio"

    mciopen.lpstrElementName = Text1.Text


Preparo la estructura con la información basica.

    resultado = mciSendCommand(0, MCI_OPEN, MCI_OPEN_ALIAS Or MCI_OPEN_TYPE Or MCI_OPEN_ELEMENT, mciopen)


Especifico las constantes necesarias en “wParam1”, para que la función tome en cuenta los valores de la estructura.

    BuscarError resultado

    mciset.dwTimeFormat = MCI_FORMAT_MILLISECONDS

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_SET, MCI_SET_TIME_FORMAT, 0)

    BuscarError resultado


Configuro el formato del tiempo, para que sea en milisegundos-

    mcistatus.dwItem = MCI_STATUS_LENGTH

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_STATUS, MCI_STATUS_ITEM, mcistatus)


Recupero la longitud, en el formato configurado arriba, es decir, recupero la cantidad de milisegundos que posee el archivo.

    BuscarError resultado

    If mcistatus.dwReturn <> 0 Then

        ProgressBar1.Min = 1

        ProgressBar1.Max = mcistatus.dwReturn

        ProgressBar1.Value = 1

        Slider1.Min = 1

        Slider1.Max = mcistatus.dwReturn

        Slider1.Value = 1

        Label2.Caption = formatotiempo(mcistatus.dwReturn)

    End If

End Sub

Botón “play”:

    Timer1.Enabled = True


Activo el timer.

    mciplay.dwFrom = 0


Especifico que empiece desde el milisegundo 0.

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_PLAY, MCI_FROM, mciplay)


Vean que especifico MCI_FROM para que tome en cuenta el valor que le paso en la estructura.

    BuscarError resultado

Botón “Stop”:

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_STOP, 0, 0)


Detengo la ejecucion.

    BuscarError resultado

    Timer1.Enabled = False

Botón “Pause”:

    If Command4.Caption = "Pausa" Then

        resultado = mciSendCommand(mciopen.wDeviceID, MCI_PAUSE, 0, 0)


Coloco en “pausa” la ejecución

        BuscarError resultado

        Command4.Caption = "Continuar"

    Else

        resultado = mciSendCommand(mciopen.wDeviceID, MCI_RESUME, 0, 0)


Continuo la ejecución de archivo

        BuscarError resultado

        Command4.Caption = "Pausa"

    End If

En el Timer:

    mcistatus.dwItem = MCI_STATUS_POSITION

    resultado = mciSendCommand(mciopen.wDeviceID, MCI_STATUS, MCI_STATUS_ITEM, mcistatus)


Busco la posicion actual en “milisegundo” que se esta ejecutando

    BuscarError resultado

.


.


.

¿mciSendString o mciSendCommand? Ya vimos como funciona cada función, la decisión ahora esta en como saber cual usar, y yo les digo usen con la que ustedes se sientan mas comodo, particularmente yo prefiero mciSendString aunque en teoria mciSendCommand debería de ser mas rapida, y esto se debe a que la cadena de string donde pasamos el mensaje, internamente la función la convierte en comandos, es decir, internamente trabaja como la función mciSendCommand, pero es obvio ver el ahorro del codigo y la facilidad de usar la cadena de mensajes en ves de las constantes y estructuras de la función mciSendCommand.
VIDEO AVI (MCISENDSTRING)
Ya visto la teoria, vamos al proyecto que usa dicha función. Primero dicho proyecto tendra que contener los siguientes componentes: Microsoft Common Dialog Control 6.0, Microsoft Windows Common Controls 6.0 (sp4). Con esto coponentes activados, procedemos a incluir los controles que presento a continuación:

4 Botones (Command).

2 Etiquetas (Label).

1 ProgressBar.

1 Slider 

1 Caja de Texto (TextBox)

1 Timer

1 Picture

Luego pegan este codigo en el formulario:

Private Declare Function mciSendString Lib "winmm.dll" Alias "mciSendStringA" (ByVal lpstrCommand As String, ByVal lpstrReturnString As String, ByVal uReturnLength As Long, ByVal hwndCallback As Long) As Long

Private Declare Function mciGetErrorString Lib "winmm.dll" Alias "mciGetErrorStringA" (ByVal dwError As Long, ByVal lpstrBuffer As String, ByVal uLength As Long) As Long

Dim sendstring As String

Dim resultado As Long

Dim error As String

Dim cadena As String

Dim slider As Boolean

Private Sub Command1_Click()

    With CommonDialog1

        .Filter = "Archivos *.avi | *.avi"

        .ShowOpen

        Text1.Text = .FileName

    End With

    MandarMensaje "close archivoavi"

    MandarMensaje "open " & Text1.Text & " type avivideo alias archivoavi parent " & Picture1.hWnd & " style child"

    MandarMensaje "set archivoavi time format milliseconds"

    MandarMensaje "status archivoavi length"

    If Val(Trim(cadena)) <> 0 Then

        ProgressBar1.Min = 1

        ProgressBar1.Max = Val(cadena)

        ProgressBar1.Value = 1

        Slider1.Min = 1

        Slider1.Max = Val(cadena)

        Slider1.Value = 1

        Label2.Caption = formatotiempo(Val(cadena))

    End If

End Sub

Private Sub Command2_Click()

    Timer1.Enabled = True

    MandarMensaje "play archivoavi from 0"

End Sub

Sub BuscarError(número As Long)

    error = Space(255)

    mciGetErrorString resultado, error, 255

    Label1.Caption = error

End Sub

Private Sub Command3_Click()

    MandarMensaje "stop archivoavi"

    Timer1.Enabled = False

End Sub

Private Sub Command4_Click()

    If Command4.Caption = "Pausa" Then

        MandarMensaje "pause archivoavi"

        Command4.Caption = "Continuar"

    Else

        MandarMensaje "resume archivoavi"

        Command4.Caption = "Pausa"

    End If

End Sub

Sub MandarMensaje(mensaje As String)

        cadena = Space(255)

        resultado = mciSendString(mensaje, cadena, Len(cadena), Me.hWnd)

        BuscarError resultado

End Sub

Private Sub Form_Load()

    Command1.Caption = "Archivo AVI"

    Command1.Height = 375

    Command1.Left = 3825

    Command1.Top = 180

    Command1.Width = 1185

    Command2.Caption = "Play"

    Command2.Height = 510

    Command2.Left = 90

    Command2.Top = 675

    Command2.Width = 1185

    Command3.Caption = "Stop"

    Command3.Height = 510

    Command3.Left = 1305

    Command3.Top = 675

    Command3.Width = 1185

    Command4.Caption = "Pausa"

    Command4.Height = 510

    Command4.Left = 2520

    Command4.Top = 675

    Command4.Width = 1185

    Label1.Caption = ""

    Label1.Height = 1140

    Label1.Left = 225

    Label1.Top = 2160

    Label1.Width = 7170

    Label2.Caption = ""

    Label2.Height = 195

    Label2.Left = 7155

    Label2.Top = 1890

    Label2.Width = 1185

    ProgressBar1.Height = 195

    ProgressBar1.Left = 270

    ProgressBar1.Top = 1665

    ProgressBar1.Width = 8250

    Slider1.Height = 360

    Slider1.Left = 90

    Slider1.Top = 1305

    Slider1.Width = 8520

    Text1.Height = 330

    Text1.Left = 90

    Text1.Top = 180

    Text1.Width = 3570

    Text1.Text = vbNullString

    Timer1.Enabled = False

    Timer1.Interval = 100

    Picture1.Height = 3760

    Picture1.Left = 270

    Picture1.Top = 2925

    Picture1.Width = 5200

    Picture1.AutoSize = True

    Me.Width = 8880

    Me.Height = 7495

End Sub

Private Sub Form_Unload(Cancel As Integer)

    MandarMensaje "close archivoavi"

End Sub

Private Sub Slider1_Change()

    If Slider1.Value = Slider1.Max Then

        Timer1.Enabled = False

    End If

End Sub

Private Sub Slider1_MouseDown(Button As Integer, Shift As Integer, x As Single, y As Single)

    slider = True

End Sub

Private Sub Slider1_MouseUp(Button As Integer, Shift As Integer, x As Single, y As Single)

    slider = False

    MandarMensaje "play archivoavi from " & Str(Slider1.Value) & " to " & Str(Slider1.Max)

End Sub

Private Sub Timer1_Timer()

    MandarMensaje "status archivoavi position"

    ProgressBar1.Value = Val(cadena)

    If Not slider Then

        Slider1.Value = Val(cadena)

    End If

    Label2.Caption = formatotiempo(ProgressBar1.Max - Val(cadena))

End Sub

Function formatotiempo(milisegundos As Long) As String

    Dim segundos As Integer

    Dim minutos As Integer

    segundos = milisegundos \ 1000

    minutos = segundos \ 60

    segundos = segundos Mod 60

    formatotiempo = Str(minutos) & ":" & Str(segundos)

End Function

CD AUDIO (MCISENDSTRING)
Ya visto la teoria, vamos al proyecto que usa dicha función. Primero dicho proyecto tendra que contener los siguientes componentes: Microsoft Common Dialog Control 6.0, Microsoft Windows Common Controls 6.0 (sp4). Con esto coponentes activados, procedemos a incluir los controles que presento a continuación:

5 Botones (Command).

2 Etiquetas (Label).

1 ProgressBar.

1 Slider 

1 Caja de Texto (TextBox)

1 Timer

Luego pegan este codigo en el formulario:

Private Declare Function mciSendString Lib "winmm.dll" Alias "mciSendStringA" (ByVal lpstrCommand As String, ByVal lpstrReturnString As String, ByVal uReturnLength As Long, ByVal hwndCallback As Long) As Long

Private Declare Function mciGetErrorString Lib "winmm.dll" Alias "mciGetErrorStringA" (ByVal dwError As Long, ByVal lpstrBuffer As String, ByVal uLength As Long) As Long

Dim sendstring As String

Dim resultado As Long

Dim error As String

Dim cadena As String

Dim slider As Boolean

Dim tracks As Integer

Dim pista As Integer

Private Sub Command5_Click()

    MandarMensaje "set archivocd door open"

End Sub

Private Sub Form_Load()

    Command1.Caption = "Buscar Pistas"

    Command1.Height = 375

    Command1.Left = 3825

    Command1.Top = 180

    Command1.Width = 1185

    Command2.Caption = "Play"

    Command2.Height = 510

    Command2.Left = 90

    Command2.Top = 675

    Command2.Width = 1185

    Command3.Caption = "Stop"

    Command3.Height = 510

    Command3.Left = 1305

    Command3.Top = 675

    Command3.Width = 1185

    Command4.Caption = "Pausa"

    Command4.Height = 510

    Command4.Left = 2520

    Command4.Top = 675

    Command4.Width = 1185

    Command5.Caption = "Expulsar"

    Command5.Height = 510

    Command5.Left = 3735

    Command5.Top = 675

    Command5.Width = 1185

    Command5.Enabled = False

    Label1.Caption = ""

    Label1.Height = 1140

    Label1.Left = 225

    Label1.Top = 2160

    Label1.Width = 7170

    Label2.Caption = ""

    Label2.Height = 195

    Label2.Left = 7155

    Label2.Top = 1890

    Label2.Width = 1185

    ProgressBar1.Height = 195

    ProgressBar1.Left = 270

    ProgressBar1.Top = 1665

    ProgressBar1.Width = 8250

    Slider1.Height = 360

    Slider1.Left = 90

    Slider1.Top = 1305

    Slider1.Width = 8520

    Timer1.Enabled = False

    Timer1.Interval = 100

    Me.Width = 8880

    Me.Height = 3495

End Sub

Private Sub Command1_Click()

    MandarMensaje "close archivocd"

    MandarMensaje "open G:\ type cdaudio alias archivocd"

    MandarMensaje "set archivocd time format milliseconds"

    MandarMensaje "status archivocd number of tracks"

    tracks = Val(cadena)

    Dim i As Integer

    Combo1.Clear

    For i = 1 To tracks

        MandarMensaje "status archivocd length track " & i

        Combo1.AddItem i & " - (" & formatotiempo(Val(cadena)) & ")"

    Next

    Command5.Enabled = True

End Sub

Private Sub Command2_Click()

    pista = Val(Mid(Combo1.Text, 1, InStr(1, Combo1.Text, "-") - 1))

    MandarMensaje "set archivocd time format tmsf"

    MandarMensaje "status archivocd length track " & pista

    If Val(Trim(cadena)) <> 0 Then

        ProgressBar1.Min = 0

        ProgressBar1.Max = (Val(Mid(cadena, 1, 2)) * 60) + Val(Mid(cadena, 4, 2))

        ProgressBar1.Value = 1

        Slider1.Min = 0

        Slider1.Max = (Val(Mid(cadena, 1, 2)) * 60) + Val(Mid(cadena, 4, 2))

        Slider1.Value = 1

        Label2.Caption = cadena

    End If

    MandarMensaje "play archivocd from " & pista & " to " & Str(pista + 1)

    Timer1.Enabled = True

End Sub

Sub BuscarError(número As Long)

    error = Space(255)

    mciGetErrorString resultado, error, 255

    Label1.Caption = error

End Sub

Private Sub Command3_Click()

    MandarMensaje "stop archivocd"

    Timer1.Enabled = False

End Sub

Private Sub Command4_Click()

    If Command4.Caption = "Pausa" Then

        MandarMensaje "pause archivocd"

        Command4.Caption = "Continuar"

    Else

        MandarMensaje "play archivocd"

        Command4.Caption = "Pausa"

    End If

End Sub

Sub MandarMensaje(mensaje As String)

        cadena = Space(255)

        resultado = mciSendString(mensaje, cadena, Len(cadena), 0)

        BuscarError resultado

End Sub

Private Sub Form_Unload(Cancel As Integer)

    MandarMensaje "close archivocd"

End Sub

Private Sub Timer1_Timer()

    MandarMensaje "status archivocd position"

    ProgressBar1.Value = (Val(Mid(cadena, 4, 2)) * 60) + Val(Mid(cadena, 7, 2))

    If Not slider Then

        Slider1.Value = (Val(Mid(cadena, 4, 2)) * 60) + Val(Mid(cadena, 7, 2))

    End If

    Label2.Caption = cadena

End Sub

Function formatotiempo(milisegundos As Long) As String

    Dim segundos As Integer

    Dim minutos As Integer

    segundos = milisegundos \ 1000

    minutos = segundos \ 60

    segundos = segundos Mod 60

    formatotiempo = Str(minutos) & ":" & Str(segundos)

End Function

VOLUMEN

Sabemos ahora como ejecutar archivos multimedia, pero como manipular el volumen de cada dispositivo de salida,  si abren “Volumen Control” veran varias columnas en donde pueden manipular los volúmenes de cada dispositivo. ¿Cómo hacer un programa que haga eso?.

Para este proyecto son necesarios los siguientes controles:

3 Hscroll o Scroll Horizontales

4 Label o etiquetas.

1 listbox

Necesitan tener el archivo “TLBINF32.DLL”  en el Windows/System, en caso de no tenerlo, en el zip que contiene este manual, ahí esta.

Luego pegen este codigo:

Dim valor As Long

Dim cadbinaria As String

Dim resto As Long

Dim cociente As Long

Dim resultado As Long

Dim negativo As Boolean

Private Type Vol

    vleft As Integer

    vright As Integer

End Type

Private Const AUXCAPS_CDAUDIO = 1   '  audio from internal CD-ROM drive

Private Const AUXCAPS_AUXIN = 2   '  audio from auxiliary input jacks

Private Const AUXCAPS_VOLUME = &H1          '  supports volume control

Private Const AUXCAPS_LRVOLUME = &H2          '  separate left-right volume control

Private Const MMSYSERR_NOERROR = 0

Private Const MMSYSERR_BASE = 0

Private Const MMSYSERR_BADDEVICEID = (MMSYSERR_BASE + 2)

Private Const MAXPNAMELEN = 32

Private Type AUXCAPS

       wMid As Integer

       wPid As Integer

       vDriverVersion As Long

       szPname As String * MAXPNAMELEN

       wTechnology As Integer

       dwSupport As Long

End Type

Private Declare Function auxGetNumDevs Lib "winmm.dll" () As Long

Private Declare Function auxGetDevCaps Lib "winmm.dll" Alias "auxGetDevCapsA" (ByVal uDeviceID As Long, lpCaps As AUXCAPS, ByVal uSize As Long) As Long

Private Declare Function auxSetVolume Lib "winmm.dll" (ByVal uDeviceID As Long, ByVal dwVolume As Long) As Long

Private Declare Function auxGetVolume Lib "winmm.dll" (ByVal uDeviceID As Long, ByRef lpdwVolume As Long) As Long

Private Declare Function loword Lib "TLBINF32" (ByVal DWord As Long) As Integer

Private Declare Function hiword Lib "TLBINF32" (ByVal DWord As Long) As Integer

Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" (hpvDest As Any, hpvSource As Any, ByVal cbCopy As Long)

Dim Vol As Long

Private Sub Form_Load()

    Me.AutoRedraw = True

    Me.Width = 5100

    Me.Height = 4035

    HScroll1.Height = 240

    HScroll1.Left = 2925

    HScroll1.Top = 1395

    HScroll1.Width = 1365

    HScroll1.Min = 0

    HScroll1.Max = 100

    HScroll2.Height = 240

    HScroll2.Left = 2925

    HScroll2.Top = 1890

    HScroll2.Width = 1365

    HScroll2.Min = 0

    HScroll2.Max = 100

    HScroll3.Height = 240

    HScroll3.Left = 2970

    HScroll3.Top = 2745

    HScroll3.Width = 1320

    HScroll3.Min = 0

    HScroll3.Max = 100

    Label1.Height = 285

    Label1.Left = 2970

    Label1.Top = 2430

    Label1.Width = 1230

    Label1.Caption = "Volume"

    Label2.Height = 240

    Label2.Left = 2925

    Label2.Top = 1125

    Label2.Width = 1365

    Label2.Caption = "Izquierdo"

    Label3.Height = 240

    Label3.Left = 2925

    Label3.Top = 1665

    Label3.Width = 1365

    Label3.Caption = "Derecho"

    label4.Height = 240

    label4.Left = 2970

    label4.Top = 765

    label4.Width = 1365

    label4.Caption = "Balance"

    List1.Height = 2400

    List1.Left = 45

    List1.Top = 630

    List1.Width = 2760

    Dim ac As AUXCAPS

    For Cnt = 0 To auxGetNumDevs - 1

        auxGetDevCaps Cnt, ac, Len(ac)

       List1.AddItem Left(ac.szPname, InStr(ac.szPname, vbNullChar) - 1)

    Next

End Sub

Private Sub List1_Click()

    Dim leftV As Long, rightV As Long

    Dim valorright As Long, valorleft As Long

    auxGetVolume List1.ListIndex, Vol

    leftV = hiword(Abs(Vol))

    rightV = loword(Abs(Vol))

    valorleft = (leftV * 100) \ 65535

    valorright = (rightV * 100) \ 65535

    If valorleft > valorright Then

        HScroll3.Value = valorleft

        HScroll1.Value = 100

        HScroll2.Value = valorright

    End If

    If valorright > valorleft Then

        HScroll3.Value = valorright

        HScroll1.Value = valorleft

        HScroll2.Value = 100

    End If

    If valorright = valorleft Then

        HScroll3.Value = valorright

        HScroll1.Value = 100

        HScroll2.Value = 100

    End If

End Sub

Private Sub HScroll1_scroll()

    CalculoVolumen

End Sub

Private Sub HScroll1_Change()

    CalculoVolumen

End Sub

Private Sub HScroll2_scroll()

    CalculoVolumen

End Sub

Private Sub HScroll2_Change()

    CalculoVolumen

End Sub

Private Sub HScroll3_scroll()

    CalculoVolumen

End Sub

Private Sub HScroll3_Change()

    CalculoVolumen

End Sub

Sub CalculoVolumen()

    Dim vTotal As Long

    Dim vright As Long, vleft As Long

    Dim v As Vol

    v.vleft = (HScroll2.Value * HScroll3.Value) / 100

    v.vright = (HScroll1.Value * HScroll3.Value) / 100

    vleft = (v.vleft * 65535) / 100

    vright = (v.vright * 65535) / 100

    If vright > 32767 Then

        v.vright = vright - 65536

    Else

        v.vright = vright

    End If

    If vleft > 32767 Then

        v.vleft = vleft - 65536

    Else

        v.vleft = vleft

    End If

    CopyMemory vTotal, v.vleft, Len(v)

    auxSetVolume List1.ListIndex, vTotal

End Sub

Cuando arranquen verán un listbox, que debera de contener todos los dispositivos de salida que poseen, al darle clic a cada dispositivo se configuran los Scroll dependiendo del volumen que tengan en ese momento, si alteran los Scroll con una música activada verán que altera el volumen.

Antes de explicar las funciones que intervienen en ello, vamos a dejar en claro, para Windows no existe un volumen MASTER, es decir, el Hscroll3 que representa un volumen global, dicho volumen Windows no lo maneja de esa manera, para Windows existe 2 volumenes derecho “right” e izquierdo “left” el control hScroll3 se programa de tal manera que altere ambos volúmenes. Ya veremos eso con mas detalle.

La primera función a estudiar sera la que no spermite saber cuantos dispositivos de salida tenemos.

auxGetNumDevs

Api: Declare Function auxGetNumDevs Lib "winmm.dll" () As Long

Esta función se encarga de retornar la cantidad de dispositivos que posee configurado Windows. Dicha cantidad se encuentra en el valor que la función devuelve.

auxGetDevCaps

Api: Declare Function auxGetDevCaps Lib "winmm.dll" Alias "auxGetDevCapsA" (ByVal uDeviceID As Long, lpCaps As AUXCAPS, ByVal uSize As Long) As Long

Esta función toma información, de un dispositivo de audio, dicha información se almacena en la estructura AUXCAPS. Como primer parámetro pasamos el ID del dispositivo a consultar, como segundo parámetro “lpCaps” le pasamos la estructura que almacenara la información, y como último parámetro le pasamos el tamaño de la estructura.

Estructura:

Private Type AUXCAPS

       wMid As Integer ‘ Identificador del fabricante

       wPid As Integer ‘ Identificador del producto

       vDriverVersion As Long ‘  version del driver

       szPname As String * MAXPNAMELEN ‘  Nombre del dispositivo

       wTechnology As Integer ‘ Tipo de salida.

       dwSupport As Long ‘ Establece si soporta ambos volumenes Izquierdo y derecho.

End Type

Particularmente solo vamos a usar “szPname”.

auxGetVolume

Api: Declare Function auxGetVolume Lib "winmm.dll" (ByVal uDeviceID As Long, ByRef lpdwVolume As Long) As Long

Con esta función tomamos el valor del volumen de cierto dispositivo, el dispositivo es indicado en el primer parametro “uDeviceID”, en el segundo parametro le pasamos una variable de tipo “long” donde se alamacena el volumen.

Dicho volumen tiene algo peculiar, ya que mencione que el volumen se divide en 2 partes derecha y parte izquierda. La parte alta de la variable “lpdwVolume” posee el volumen de la salida izquierda y la parte baja posee el volumen de la parte derecha. Esto significa que si la variable “lpdwVolume” es de 32 bits, cada volumen es de 16 Bits, con esto deducimos que el volumen de cada salida va desde 0 hasta 65535.

auxSetVolume

Api: Declare Function auxSetVolume Lib "winmm.dll" (ByVal uDeviceID As Long, ByVal dwVolume As Long) As Long

Esta función le asigna un nuevo volumen a un dispositivo de salida, dicho dispositivo se especifica en el primer parametro, y en el segundo parametro se especifica el nuevo volumen. ¿Pero el volumen no se divide en dos izquierda-derecha? Cierto, la parte alta de “dwVolume” contendra el nuevo volumen izquierdo y y la parte baja tendra el volumen derecho.

Seguramente creen que es sencillo manejar esto, ciertamente si estuviéramos en C o C++ sería facil, pero en VB hay que usar truquitos para hacer funcionar esto. Por ejemplo: como poner el volumen izquierdo en la parte alta y como poner el volumen derecho en la parte baja. Todo eso vamos a verlo directo con las líneas del código.

Empecemos en el LOAD:

    Dim ac As AUXCAPS

    For Cnt = 0 To auxGetNumDevs – 1 ‘  Por cada dispositivo que existe tomamos su información.

        auxGetDevCaps Cnt, ac, Len(ac) ‘
       List1.AddItem Left(ac.szPname, InStr(ac.szPname, vbNullChar) - 1)

    Next

List1_Click:

    auxGetVolume List1.ListIndex, Vol

   
Obtengo el volumen del dispositivo seleccionado

    leftV = hiword(Abs(Vol))

    rightV = loword(Abs(Vol))


Separamos los volumenes izquierdo y derechos.

    valorleft = (leftV * 100) \ 65535

    valorright = (rightV * 100) \ 65535


Lo llevamos a una escala de 100.

    If valorleft > valorright Then

        HScroll3.Value = valorleft

        HScroll1.Value = 100

        HScroll2.Value = valorright

    End If

    If valorright > valorleft Then

        HScroll3.Value = valorright

        HScroll1.Value = valorleft

        HScroll2.Value = 100

    End If

    If valorright = valorleft Then

        HScroll3.Value = valorright

        HScroll1.Value = 100

        HScroll2.Value = 100

    End If

Configuramos cada scroll con su valor correspondiente.

Función “CalculoVolumen”:

    Dim v As Vol


Creo una variable del tipo “Vol”, “Vol” es una estructura que posee 2 variables integer, dicha estructura la vamos a usar para hacer un truco.

    v.vleft = (HScroll2.Value * HScroll3.Value) / 100

    v.vright = (HScroll1.Value * HScroll3.Value) / 100


Dependiendo del volumen “general”, configuro los volumenes para cada lado, recuerden que ustedes pueden alterar el volumen de ambos al mismo tiempo por el Hscroll3, dichos cambios se reflejan con estos calculos.


    vleft = (v.vleft * 65535) / 100

    vright = (v.vright * 65535) / 100

    
Los calculos que estan en una escala de 100 los convierto a una escala desde 0 hasta 65535. Estas variables vleft y vright pueden llegar hasta 65535 ya que son del tipo LONG.

    If vright > 32767 Then

        v.vright = vright - 65536

    Else

        v.vright = vright

    End If

    If vleft > 32767 Then

        v.vleft = vleft - 65536

    Else

        v.vleft = vleft

    End If


Llevo el volumen a un formato que puede entender la función. Vean que los valores negativos se valen, ya que las función sabe interpretarlos.       

    CopyMemory vTotal, v.vleft, Len(v)


CopyMemory, es casi idéntico a la función visto en el Capítulo  de “menu” que se llamaba MoveMemory. Con esto lo que hacemos es poner de manera correcta la información en la variable vTotal. Es decir, la parte alta de Vtotal tendra el volumen izquierdo y la parte baja tendra el volumen derecho, por esa razón la estructura es así:

Private Type Vol

    vleft As Integer

    vright As Integer

End Type

    auxSetVolume List1.ListIndex, vTotal

  
Configuramos el nuevo volumen.

End Sub

Con esto finalizamos la parte de multimedia, existen muchas funciones que manipulan los diferentes dispositivos multimedia, pero creo que cubri las mas importantes.

Capítulo  VIII.  Cadenas (String) y Memoria

En este Capítulo  vamos a dar una breve introducción, de 2 temas que seguramente será de muy poco uso en VB, pero que algunas veces es necesario conocerlos para el manejo de algunas funciones API’S, por ejemplo, las funciones que vamos a ver a nivel de Cadenas o String, no ve ser necesario usarlas en VB, ya que dicho lenguaje presenta funciones que hacen el trabajo. Pero uno nunca sabe cuando uno puede usar estas funciones, inclusive para aquellos osados que programen en C entonces si necesitaran el conocer estas funciones.

CADENAS O STRING
En esta sección veremos algunas funciones API’S utilizadas para la manipulación de caracteres, para este proyecto es necesario: 3 Label o etiquetas, 1 listBox y 1 caja de texto o TextBox.

Luego peguen este código:

Private Declare Function CharLower Lib "user32" Alias "CharLowerA" (ByVal lpsz As String) As String

Private Declare Function CharLowerBuff Lib "user32" Alias "CharLowerBuffA" (ByVal lpsz As String, ByVal cchLength As Long) As Long

Private Declare Function CharUpper Lib "user32" Alias "CharUpperA" (ByVal lpsz As String) As String

Private Declare Function CharUpperBuff Lib "user32" Alias "CharUpperBuffA" (ByVal lpsz As String, ByVal cchLength As Long) As Long

Private Declare Function IsCharAlpha Lib "user32" Alias "IsCharAlphaA" (ByVal cChar As Byte) As Long

Private Declare Function IsCharAlphaNumeric Lib "user32" Alias "IsCharAlphaNumericA" (ByVal cChar As Byte) As Long

Private Declare Function IsCharLower Lib "user32" Alias "IsCharLowerA" (ByVal cChar As Byte) As Long

Private Declare Function IsCharUpper Lib "user32" Alias "IsCharUpperA" (ByVal cChar As Byte) As Long

Private Declare Function lstrcat Lib "kernel32" Alias "lstrcatA" (ByVal lpString1 As String, ByVal lpString2 As String) As Long

Private Declare Function lstrcmp Lib "kernel32" Alias "lstrcmpA" (ByVal lpString1 As String, ByVal lpString2 As String) As Long

Private Declare Function lstrcmpi Lib "kernel32" Alias "lstrcmpiA" (ByVal lpString1 As String, ByVal lpString2 As String) As Long

Private Declare Function lstrcpy Lib "kernel32" Alias "lstrcpyA" (ByVal lpString1 As String, ByVal lpString2 As String) As Long

Private Declare Function lstrcpyn Lib "kernel32" Alias "lstrcpynA" (ByVal lpString1 As String, ByVal lpString2 As String, ByVal iMaxLength As Long) As Long

Private Declare Function lstrlen Lib "kernel32" Alias "lstrlenA" (ByVal lpString As String) As Long

Dim s1 As String, s2 As String, s3 As String

Private Sub Form_Load()

    Me.Width = 5880

    Me.Height = 2475

    Label1.Height = 285

    Label1.Left = 2340

    Label1.Top = 90

    Label1.Width = 2715

    Label2.Height = 240

    Label2.Left = 2340

    Label2.Top = 315

    Label2.Width = 3210

    Label3.Height = 285

    Label3.Left = 2340

    Label3.Top = 810

    Label3.Width = 2715

    List1.Height = 1815

    List1.Left = 90

    List1.Top = 90

    List1.Width = 2130

    Text1.Height = 330

    Text1.Left = 2340

    Text1.Top = 1035

    Text1.Width = 2850

    Text1.Text = vbNullString

    List1.AddItem "CharLower"

    List1.AddItem "CharLowerBuff"

    List1.AddItem "CharUpper"

    List1.AddItem "CharUpperBuff"

    List1.AddItem "IsCharAlpha"

    List1.AddItem "IsCharAlphaNumeric"

    List1.AddItem "IsCharAlphaLower"

    List1.AddItem "IsCharAlphaUpper"

    List1.AddItem "lstrcat"

    List1.AddItem "lstrcmp"

    List1.AddItem "lstrcmpi"

    List1.AddItem "lstrcpy"

    List1.AddItem "lstrcpyn"

    List1.AddItem "lstrlen"

End Sub

Private Sub List1_Click()

    Select Case List1.ListIndex

    Case 0

        s1 = "HOLA!": Label2 = s1

        s2 = CharLower(s1): Text1.Text = s2

    Case 1

        s1 = "HOLA!": Label2 = s1

        CharLowerBuff s1, 2: Text1.Text = s1

    Case 2

        s1 = "hola!": Label2 = s1

        s2 = CharUpper(s1): Text1.Text = s2

    Case 3

        s1 = "hola!": Label2 = s1

        CharUpperBuff s1, 2: Text1.Text = s1

    Case 4

        Dim ascii As Byte

        ascii = 65: Label2 = Chr(ascii)

        If IsCharAlpha(ascii) Then

            Text1.Text = True

        Else

            Text1.Text = False

        End If

    Case 5

        Dim ascii2 As Byte

        ascii2 = 65: Label2 = Chr(ascii2)

        If IsCharAlphaNumeric(ascii2) Then

            Text1.Text = True

        Else

            Text1.Text = False

        End If

    Case 6

        Dim ascii3 As Byte

        ascii3 = 65: Label2 = Chr(ascii3)

        If IsCharLower(ascii3) Then

            Text1.Text = True

        Else

            Text1.Text = False

        End If

    Case 7

        Dim ascii4 As Byte

        ascii4 = 65: Label2 = Chr(ascii4)

        If IsCharUpper(ascii4) Then

            Text1.Text = True

        Else

            Text1.Text = False

        End If

    Case 8

        s1 = "String 1 & "

        s2 = "String 2"

        lstrcat s1, s2               

        Text1.Text = s1

    Case 9

        s1 = "String 1"

        s2 = "string 1"

        If lstrcmp(s1, s2) = 0 Then

            Text1.Text = "Son iguales"

        Else

            Text1.Text = "No son iguales"

        End If

    Case 10

        s1 = "String 1"

        s2 = "string 1"

        If lstrcmpi(s1, s2) = 0 Then

            Text1.Text = "Son iguales"

        Else

            Text1.Text = "No son iguales"

        End If

    Case 11

        s1 = "String 1"

        s2 = Space(lstrlen(s1))

        lstrcpy s2, s1

        Text1.Text = s2

    Case 12

        s1 = "String 1"

        s2 = Space(4)

        lstrcpyn s2, s1, 4

        Text1.Text = s2

    Case 13

        s1 = "String 1"

        Text1.Text = lstrlen(s1)

    End Select

End Sub

CharLower

Api: Declare Function CharLower Lib "user32" Alias "CharLowerA" (ByVal lpsz As String) As String

Función encargada de poner en minúscula una cadena de texto, dicha cadena se le pasa como único parámetro, y la función retorna la nueva cadena con todos los caracteres en minúscula.

CharLowerBuff

Api: Declare Function CharLowerBuff Lib "user32" Alias "CharLowerBuffA" (ByVal lpsz As String, ByVal cchLength As Long) As Long

Hace exactamente lo mismo que la función descrita arriba, con la diferencia que como segundo parámetro le pasamos la cantidad de caracteres que queramos poner en minúscula, es decir, con esta función podemos convertir en minúscula una porción de la cadena de caracteres.

CharUpper

Api: Declare Function CharUpper Lib "user32" Alias "CharUpperA" (ByVal lpsz As String) As String

Igual que la función CharLower con la diferencia que convierte los caracteres que están en minúscula en mayúsculas.

CharUpperBuff

Api: Declare Function CharUpperBuff Lib "user32" Alias "CharUpperBuffA" (ByVal lpsz As String, ByVal cchLength As Long) As Long

Igual que la función CharUpperBuff con la diferencia que convierte los caracteres que están en minúscula en mayúsculas.

IsCharAlpha

Api: Declare Function IsCharAlpha Lib "user32" Alias "IsCharAlphaA" (ByVal cChar As Byte) As Long

Determina si un carácter es “Alpha”, es decir, es cualquier letra del abecedario, ya sea en mayúscula  o en minúscula, dicho carácter, se le pasa como primer parámetro en forma de “byte” lo cual significa que se le pasa el código ASCII. Si el carácter es del tipo “Alpha” entonces retornara un valor distinto de 0. En caso contrario retorna 0.

IsCharAlphaNumeric

Api: Declare Function IsCharAlphaNumeric Lib "user32" Alias "IsCharAlphaNumericA" (ByVal cChar As Byte) As Long

Determina si un carácter es del tipo “Alpha” numérico, es decir, todos los caracteres del abecedario incluyendo los números. Las condiciones del primer parámetro son iguales a IsCharAlpha.

IsCharLower
Api: Declare Function IsCharLower Lib "user32" Alias "IsCharLowerA" (ByVal cChar As Byte) As Long

Determina si un carácter esta en minúscula, si el carácter esta en minúscula la función retorna un valor distinto de 0.

IsCharUpper
Api: Declare Function IsCharUpper Lib "user32" Alias "IsCharUpperA" (ByVal cChar As Byte) As Long

Determina si un carácter esta en mayúscula, si el carácter esta en mayúscula la función retorna un valor distinto de 0.

lstrcat
Api: Declare Function lstrcat Lib "kernel32" Alias "lstrcatA" (ByVal lpString1 As String, ByVal lpString2 As String) As Long

Función que concatena 2 cadenas, es decir, pega “lpString2” en “lpString1”, ejm:


Cadena1 = “Hola”


Cadena2 = “ Mundo”.

Cadena1 después de lstrcat = “Hola Mundo”

lstrcmp

Private Declare Function lstrcmp Lib "kernel32" Alias "lstrcmpA" (ByVal lpString1 As String, ByVal lpString2 As String) As Long

Compara dos cadenas de caracteres, dicha comparación “es sensitiva a mayúsculas y minúsculas”, es decir, para esta función:

s1 = "String 1"

s2 = "string 1"

Estas cadenas, no son iguales, ya que “s1” tiene “S” y “s2” tiene “s”.

lstrcmpi

Api: Declare Function lstrcmpi Lib "kernel32" Alias "lstrcmpiA" (ByVal lpString1 As String, ByVal lpString2 As String) As Long

Hace lo mismo que la anterior con la diferencia que “no es sensitiva”, es decir, para esta función:

s1 = "String 1"

s2 = "string 1"

Son iguales.

lstrcpy

Private Declare Function lstrcpy Lib "kernel32" Alias "lstrcpyA" (ByVal lpString1 As String, ByVal lpString2 As String) As Long

Copia la cadena que hay en “lpString2” en “lpString1”.

lstrcpyn
Private Declare Function lstrcpyn Lib "kernel32" Alias "lstrcpynA" (ByVal lpString1 As String, ByVal lpString2 As String, ByVal iMaxLength As Long) As Long

Copia una cantidad de bytes específicos que hay en “lpString2” en “lpString1”. Dicha cantidad de bytes es pasada en el tercer parámetro de la función.

lstrlen

Api: Declare Function lstrlen Lib "kernel32" Alias "lstrlenA" (ByVal lpString As String) As Long

Determina la longitud de la cadena de texto, pero dicha longitud dependerá del formato del texto, el valor que retorna será la cantidad en bytes, si la cadena es del tipo (ANSI) o retornara la cantidad de caracteres si el del tipo (UNICODE).

MEMORIA

En esta sección vamos a tratar las funciones básicas para tratar con la memoria del sistema, todos sabemos que con la introducción de Windows 95, se añadió lo que se llama memoria virtual, dicha memoria esta localizada en el disco duro, y el sistema la usa cuando la memoria ram se encuentra llena, en el proyecto que veremos a continuación trabajamos con la memoria RAM, pero las explicaciones de las funciones que veremos son análogas para el manejo de la memoria virtual.

Para este proyecto es necesario: 2 botones, y 2 etiquetas o label. Luego pegan este código:

Private Const GMEM_FIXED = &H0

Private Const GMEM_MOVEABLE = &H2

Private Const GMEM_DDESHARE = &H2000

Private Const GMEM_SHARE = &H2000

Private Const GMEM_DISCARDABLE = &H100

Private Const GMEM_NOT_BANKED = &H1000

Private Const GMEM_LOWER = GMEM_NOT_BANKED

Private Const GMEM_NOCOMPACT = &H10

Private Const GMEM_NODISCARD = &H20

Private Const GMEM_NOTIFY = &H4000

Private Const GMEM_ZEROINIT = &H40

Private Declare Function GlobalAlloc Lib "kernel32" (ByVal wFlags As Long, ByVal dwBytes As Long) As Long

Private Declare Function GlobalReAlloc Lib "kernel32" (ByVal hMem As Long, ByVal dwBytes As Long, ByVal wFlags As Long) As Long

Private Declare Function GlobalFree Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Function GlobalLock Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Function GlobalUnlock Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Function VirtualAlloc Lib "kernel32" (lpAddress As Any, ByVal dwSize As Long, ByVal flAllocationType As Long, ByVal flProtect As Long) As Long

Private Declare Function VirtualLock Lib "kernel32" (lpAddress As Any, ByVal dwSize As Long) As Long

Private Declare Function VirtualUnlock Lib "kernel32" (lpAddress As Any, ByVal dwSize As Long) As Long

Private Declare Function VirtualFree Lib "kernel32" (lpAddress As Any, ByVal dwSize As Long, ByVal dwFreeType As Long) As Long

Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" (Destination As Any, Source As Any, ByVal Length As Long)

Private Declare Sub FillMemory Lib "kernel32.dll" Alias "RtlFillMemory" (Destination As Any, ByVal Length As Long, ByVal Fill As Byte)

Private Declare Function lstrlen Lib "kernel32" Alias "lstrlenA" (ByVal lpString As Any) As Long

Dim s1 As String, s2 As String

Private Sub Command1_Click()

    Dim espacio As Long

    Dim espaciofijo As Long

    Dim num As Long, num2 As Long

    num = 100

    espacio = GlobalAlloc(GMEM_ZEROINIT, 4)

    espaciofijo = GlobalLock(espacio)

    CopyMemory ByVal espaciofijo, num, 4

    CopyMemory num2, ByVal espaciofijo, 4

    Label2 = num2

    GlobalUnlock espaciofijo

    GlobalFree espacio

End Sub

Private Sub Command2_Click()

    Dim espacio As Long

    Dim espaciofijo As Long

    s1 = "Hola mundo!"

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1))

    CopyMemory ByVal espacio, ByVal s1, Len(s1)

    s2 = String$(lstrlen(espacio), 0)

    CopyMemory ByVal s2, ByVal espacio, lstrlen(s1)

    Label1 = s2

    GlobalFree espacio

End Sub

Private Sub Form_Load()

    Me.Width = 3255

    Me.Height = 1215    

    Command1.Height = 465

    Command1.Left = 135

    Command1.Top = 90

    Command1.Width = 1230    

    Command2.Height = 465

    Command2.Left = 1710

    Command2.Top = 90

    Command2.Width = 1230    

    Label1.Height = 465

    Label1.Left = 1710

    Label1.Top = 585

    Label1.Width = 1230    

    Label2.Height = 465

    Label2.Left = 135

    Label2.Top = 585

    Label2.Width = 1230

End Sub

Cuando arranquen y aprieten los botones, verán que las etiquetas se llenan con información, ustedes podrían preguntarse, ¿Para que hacer todo esto, si con solo asignar variable1=variable2, estamos listo?. Ciertamente, esto es un proceso un poco engorroso, pero algunas veces, para manipular algunas API’S es necesario tener en cuenta estas funciones primordialmente CopyMemory.

GlobalAlloc

Api: Declare Function GlobalAlloc Lib "kernel32" (ByVal wFlags As Long, ByVal dwBytes As Long) As Long

Función encargada de reservar un espacio en memoria, el primer parámetro puede tener una combinación OR de las siguientes constantes:

Private Const GMEM_FIXED = &H0 ‘ Reserva un espacio ya fijo, en la memoria.

Private Const GMEM_MOVEABLE = &H2 ‘ Reserva un especio el cual no esta fijo, es decir, Windows, tiene la libertad de moverlo.

Private Const GMEM_DDESHARE = &H2000 
Private Const GMEM_SHARE = &H2000  ‘ Estas dos constantes, sirven para reservar un especio de memoria compartido.

Private Const GMEM_DISCARDABLE = &H100 
Private Const GMEM_NOT_BANKED = &H1000 ‘ 

Private Const GMEM_LOWER = GMEM_NOT_BANKED

Private Const GMEM_NOTIFY = &H4000 

Private Const GMEM_NOCOMPACT = &H10

Private Const GMEM_NODISCARD = &H20 ‘ Todas estas constante son ignorada por el sistema, solo se usa para mantener compatibilidad con aplicaciones de 16 Bits.

Private Const GMEM_ZEROINIT = &H40 ‘ Inicializa el contenido de la memoria con 0.

Como segundo parámetro “dwBytes”, le pasamos la cantidad de bytes que queremos reservar para dicha memoria.

En caso de no poseer suficiente espacio en la pila de memoria, la función retornara NULL, de lo contrario retorna el “handle” o identificador de la memoria, en caso de haber usado GMEM_FIXED la función retorna directamente el puntero de la memoria, es decir, la ubicación en donde almacenaremos nuestra información.

GlobalReAlloc

Api: Declare Function GlobalReAlloc Lib "kernel32" (ByVal hMem As Long, ByVal dwBytes As Long, ByVal wFlags As Long) As Long

Función que sirve para cambiar los atributos o tamaño a un “handle” de memoria ya creado.  Como primer parámetro le pasamos el “handle” de la memoria, como segundo parámetro le pasamos el nuevo tamaño en bytes, y como tercer parámetro les pasamos los “flag” los cuales son los mismos que las constantes de la función GloballAlloc.

GlobalLock

Api: Declare Function GlobalLock Lib "kernel32" (ByVal hMem As Long) As Long

Función que bloquea el espacio de memoria asignado con GlobalAlloc, ¿Qué quiere decir bloquear?. Cuando creamos un espacio en memoria, sin usar la constante GMEM_FIXED, el sistema operativo esta en la autoridad de poder mover dicha información, esto puede ocasionar fallas en el programa. Imaginemos que tenemos una variable “a”, que esta en la posición 1 de la memoria , y le asignamos el valor de 5, si creamos la variable “a” con GlobalAlloc sin haber usado GlobalLock, cuando vayamos a consultar la variable “a” puede que Windows nos retorne un error, y esto se debe a que Windows pudo haber cambiado la posición en donde estaba la información de “a”, con GloballLock, nosotros lo que hacemos es fijar o bloquear ese espacio para que Windows no lo mueva y así poder trabajar dicho espacio sin problemas.

El único parámetro a usar es el “handle” de la memoria que nos retorna GlobalAlloc.
GlobalUnlock

Api: Declare Function GlobalUnlock Lib "kernel32" (ByVal hMem As Long) As Long

Lo que hace es desbloquear un “handle” de memoria.

GlobalFree

Api: Declare Function GlobalFree Lib "kernel32" (ByVal hMem As Long) As Long

Con esto liberamos el espacio de memoria que fue utilizado, dejando que otro programa pueda usar dicho espacio. Todo espacio que creamos con GlobalAlloc tiene que ser liberado, sino aunque cerremos el programa que manipulaba la memoria la información sigue estando contenida en la memoria RAM de la computadora.

CopyMemory

Api: Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" (Destination As Any, Source As Any, ByVal Length As Long)

Función encargada de copiar la información de un espacio de memoria a otro. Los dos primeros parámetro hablan por si solo, y el tercer parámetro es la longitud en bytes de la cantidad de información que queremos copiar.

FillMemory

Private Declare Sub FillMemory Lib "kernel32.dll" Alias "RtlFillMemory" (Destination As Any, ByVal Length As Long, ByVal Fill As Byte)

Llena el espacio en memoria, con la información que se le pasa en el tercer parámetro “fill”, es decir:


Fillmemory(espacio,4,65)

Aquí lo que estoy haciendo es que en la dirección de memoria contenido en la variable “espacio”, almacena 4 bytes, donde cada byte tendrá la letra “A”, si lo vemos como carácter o 65 si lo vemos como número. ¿Lo ven?. Llena el contenido de la memoria con la información pasada en “fill”.

Vayamos a los ejemplos, primero veamos el “command1”:

    Dim espacio As Long

    Dim espaciofijo As Long ‘ Declaro las dos variable que almacenaran la dirección de memoria reservada.

    Dim num As Long, num2 As Long

    num = 100

    espacio = GlobalAlloc(GMEM_ZEROINIT, 4) ‘ Reservo 4 bytes de memoria, y los inicializo en 0

    espaciofijo = GlobalLock(espacio)  ‘ Fijo el espacio para que Windows no cambie la información.

    CopyMemory ByVal espaciofijo, num, 4 ‘ Copio el valor de “num”, en la dirección de memoria de “espaciofijo”.

    CopyMemory num2, ByVal espaciofijo, 4 ‘ Copio lo que hay en la dirección “espaciofijo” en la variable “num2”

    Label2 = num2 ‘ Muestro el contenido “num2” que es igual al de “num1”

    GlobalUnlock espaciofijo ‘ Desbloqueé el espacio que esta bloqueado, para liberarlo.

    GlobalFree espacio ‘ Libero el espacio.

Dense cuenta de algo, la variable “espaciofijo” contiene la dirección de memoria donde almacenaremos información, es decir:

CopyMemory ByVal espaciofijo, num, 4

Es incorrecto pensar que “espaciofijo” ahora tiene el valor de “num”, NO!, la dirección que apunta “espaciofijo” contiene el valor de “num”, que es algo muy distinto. “espaciofijo” es una variable ubicado en X lugar de la memoria, cuyo numero de tipo “long” que contiene específicamente una dirección en la memoria.  Espero que vean la diferencia. Para los conocedores de C seguro no habrá problema.

Vayamos al “command2”:

   Dim espacio As Long               

    s1 = "Hola mundo!"

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1)) ‘ Reservamos un espacio ya fijo en la memoria, esto es gracias a la constante GMEM_FIXED es decir, no es necesario el uso de la función GlobalLock.

    CopyMemory ByVal espacio, ByVal s1, Len(s1) ‘ Copiamos la cadena de caracteres “s1”, en la dirección de memoria que tiene almacenada la variable “espacio”.

    s2 = String$(lstrlen(espacio), 0) ‘ Inicializamos el espacio de memoria donde almacenáremos el contenido de la dirección de memoria almacenada en la variable “espacio”.

    CopyMemory ByVal s2, ByVal espacio, lstrlen(s1) ‘ Copiamos el contenido de la dirección que esta en “espacio” en la variable “s2”.

    Label1 = s2

    GlobalFree espacio ‘Liberamos dicho espacio.

Como pueden ver no es muy cómodo trabajar de esta manera, y mas cuando todo se pudo haber solucionado con s1 = s2, y listo!, pero resulta que algunas veces las funciones API’S de Windows nos regresan punteros de memoria que para visualizar su contenido tenemos que hacer lo antes expuesto arriba, no tanto el procedimiento como tal, sino el uso de CopyMemory.

Funciones “Virtual”.

VirtualAlloc

Api: Declare Function VirtualAlloc Lib "kernel32" (lpAddress As Any, ByVal dwSize As Long, ByVal flAllocationType As Long, ByVal flProtect As Long) As Long

VirtualLock

Api: Declare Function VirtualLock Lib "kernel32" (lpAddress As Any, ByVal dwSize As Long) As Long

VirtualUnlock

Api: Declare Function VirtualUnlock Lib "kernel32" (lpAddress As Any, ByVal dwSize As Long) As Long

VirtualFree

Api: Declare Function VirtualFree Lib "kernel32" (lpAddress As Any, ByVal dwSize As Long, ByVal dwFreeType As Long) As Long

Estas funciones trabajan exactamente igual a sus homologas de arriba. Con la diferencia que trabajan sobre la memoria virtual.

Capítulo  IX.  Archivos

En este Capítulo  vamos a cubrir las funciones encargadas de la manipulación de archivos, a la vez vamos a ver funciones que nos permite obtener información de las unidades físicas de disco y los propios archivos. Es necesario que presten mucha atención ya que el manejo de archivo con las funciones API’S Win32 no es tan sencillo como creen, esto lo digo ya que VB proporciona controles y librerías que hacen el manejo de archivo rápido y sencillo pero ahora veremos lo que hay detrás de dichos controles.

CREANDO Y LEYENDO ARCHIVOS

Para este proyecto vamos a necesitar un formulario, que debe de contener 10 botones “command”, 1 label o etiqueta, 1 ProgressBar o barra de progreso y 1 CommonDialog. Luego pegan este código:

Private Const MAX_PATH = 260

Private Const OFS_MAXPATHNAME = 128

Private Const INVALID_HANDLE_VALUE = -1

Private Const GENERIC_READ = &H80000000

Private Const GENERIC_WRITE = &H40000000

Private Const FILE_SHARE_DELETE = &H4

Private Const FILE_SHARE_READ = &H1

Private Const FILE_SHARE_WRITE = &H2

Private Const CREATE_NEW = 1

Private Const CREATE_ALWAYS = 2

Private Const OPEN_EXISTING = 3

Private Const OPEN_ALWAYS = 4

Private Const TRUNCATE_EXISTING = 5

Private Const FILE_ATTRIBUTE_ARCHIVE = &H20

Private Const FILE_ATTRIBUTE_HIDDEN = &H2

Private Const FILE_ATTRIBUTE_NORMAL = &H80

Private Const FILE_ATTRIBUTE_OFFLINE = &H1000

Private Const FILE_ATTRIBUTE_READONLY = &H1

Private Const FILE_ATTRIBUTE_SYSTEM = &H4

Private Const FILE_ATTRIBUTE_TEMPORARY = &H100

Private Const FILE_FLAG_WRITE_THROUGH = &H80000000

Private Const FILE_FLAG_OVERLAPPED = &H40000000

Private Const FILE_FLAG_NO_BUFFERING = &H20000000

Private Const FILE_FLAG_RANDOM_ACCESS = &H10000000

Private Const FILE_FLAG_SEQUENTIAL_SCAN = &H8000000

Private Const FILE_FLAG_DELETE_ON_CLOSE = &H4000000

Private Const FILE_FLAG_BACKUP_SEMANTICS = &H2000000

Private Const FILE_FLAG_POSIX_SEMANTICS = &H1000000

Private Const FILE_FLAG_OPEN_REPARSE_POINT = &H200000

Private Const FILE_FLAG_OPEN_NO_RECALL = &H100000

'Private Const SECURITY_ANONYMOUS = (SecurityAnonymous << 16 )

'Private Const SECURITY_IDENTIFICATION = (SecurityIdentification << 16 )

'Private Const SECURITY_IMPERSONATION = (SecurityImpersonation << 16 )

'Private Const SECURITY_DELEGATION = (SecurityDelegation << 16 )

Private Const SECURITY_CONTEXT_TRACKING = &H40000

Private Const SECURITY_EFFECTIVE_ONLY = &H80000

Private Const FILE_BEGIN = 0

Private Const FILE_END = 2

'Private Declare Function CreateDirectory Lib "kernel32" Alias "CreateDirectoryA" (ByVal lpPathName As String, lpSecurityAttributes As SECURITY_ATTRIBUTES) As Long

'Private Declare Function CreateDirectoryEx Lib "kernel32" Alias "CreateDirectoryExA" (ByVal lpTemplateDirectory As String, ByVal lpNewDirectory As String, lpSecurityAttributes As SECURITY_ATTRIBUTES) As Long

Private Declare Function CreateDirectory Lib "kernel32" Alias "CreateDirectoryA" (ByVal lpPathName As String, lpSecurityAttributes As Long) As Long

Private Declare Function CreateDirectoryEx Lib "kernel32" Alias "CreateDirectoryExA" (ByVal lpTemplateDirectory As String, ByVal lpNewDirectory As String, lpSecurityAttributes As Long) As Long

'Private Declare Function CreateFile Lib "kernel32" Alias "CreateFileA" (ByVal lpFileName As String, ByVal dwDesiredAccess As Long, ByVal dwShareMode As Long, lpSecurityAttributes As SECURITY_ATTRIBUTES, ByVal dwCreationDisposition As Long, ByVal dwFlagsAndAttributes As Long, ByVal hTemplateFile As Long) As Long

Private Declare Function CreateFile Lib "kernel32" Alias "CreateFileA" (ByVal lpFileName As String, ByVal dwDesiredAccess As Long, ByVal dwShareMode As Long, lpSecurityAttributes As Long, ByVal dwCreationDisposition As Long, ByVal dwFlagsAndAttributes As Long, ByVal hTemplateFile As Long) As Long

Private Declare Function DeleteFile Lib "kernel32" Alias "DeleteFileA" (ByVal lpFileName As String) As Long

Private Declare Function GetFileAttributes Lib "kernel32" Alias "GetFileAttributesA" (ByVal lpFileName As String) As Long

'Private Declare Function ReadFile Lib "kernel32" (ByVal hFile As Long, lpBuffer As Any, ByVal nNumberOfBytesToRead As Long, lpNumberOfBytesRead As Long, lpOverlapped As OVERLAPPED) As Long

Private Declare Function ReadFile Lib "kernel32" (ByVal hFile As Long, lpBuffer As Any, ByVal nNumberOfBytesToRead As Long, lpNumberOfBytesRead As Long, lpOverlapped As Long) As Long

Private Declare Function RemoveDirectory Lib "kernel32" Alias "RemoveDirectoryA" (ByVal lpPathName As String) As Long

Private Declare Function SetFilePointer Lib "kernel32" (ByVal hFile As Long, ByVal lDistanceToMove As Long, lpDistanceToMoveHigh As Long, ByVal dwMoveMethod As Long) As Long

''Private Declare Function WriteFile Lib "kernel32" (ByVal hFile As Long, lpBuffer As Any, ByVal nNumberOfBytesToWrite As Long, lpNumberOfBytesWritten As Long, lpOverlapped As OVERLAPPED) As Long

Private Declare Function WriteFile Lib "kernel32" (ByVal hFile As Long, lpBuffer As Any, ByVal nNumberOfBytesToWrite As Long, lpNumberOfBytesWritten As Long, lpOverlapped As Long) As Long

Private Declare Function GetFileSize Lib "kernel32" (ByVal hFile As Long, lpFileSizeHigh As Long) As Long

Private Declare Function CloseHandle Lib "kernel32" (ByVal hObject As Long) As Long

Private Const GMEM_FIXED = &H0

Private Const GMEM_MOVEABLE = &H2

Private Const GMEM_DDESHARE = &H2000

Private Const GMEM_SHARE = &H2000

Private Const GMEM_DISCARDABLE = &H100

Private Const GMEM_NOT_BANKED = &H1000

Private Const GMEM_LOWER = GMEM_NOT_BANKED

Private Const GMEM_NOCOMPACT = &H10

Private Const GMEM_NODISCARD = &H20

Private Const GMEM_NOTIFY = &H4000

Private Const GMEM_ZEROINIT = &H40

Private Declare Function GlobalAlloc Lib "kernel32" (ByVal wFlags As Long, ByVal dwBytes As Long) As Long

Private Declare Function GlobalReAlloc Lib "kernel32" (ByVal hMem As Long, ByVal dwBytes As Long, ByVal wFlags As Long) As Long

Private Declare Function GlobalFree Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Function GlobalUnlock Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" (pDst As Any, pSrc As Any, ByVal ByteLen As Long)

Private Declare Function lstrlen Lib "kernel32" Alias "lstrlenA" (ByVal lpString As String) As Long

Dim s1 As String

Dim bBytes() As Byte

Dim hFile As Long

Dim espacio As Long

Dim tamaño As Long

Dim cadena As String

Private Sub Command1_Click()

    hFile = CreateFile("C:\prueba.txt", GENERIC_WRITE, FILE_SHARE_READ Or FILE_SHARE_WRITE, ByVal 0&, OPEN_ALWAYS, 0, 0)

    If hFile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If

    s1 = "Hola Mundo!"

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1))

    CopyMemory ByVal espacio, ByVal s1, Len(s1)

    Dim ret As Long

    WriteFile hFile, ByVal espacio, Len(s1), ret, ByVal 0&

    GlobalFree espacio

    'SetFilePointer hFile, 0, ByVal 0&, FILE_BEGIN

    SetFilePointer hFile, 0, ByVal 0&, FILE_END

    s1 = "como estas"

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1))

    CopyMemory ByVal espacio, ByVal s1, Len(s1)

    WriteFile hFile, ByVal espacio, Len(s1), ret, ByVal 0&

    GlobalFree espacio

    CloseHandle hFile

End Sub

Private Sub Command10_Click()

    RemoveDirectory "C:\ejemplo"

    RemoveDirectory "C:\ejemplo 2"

End Sub

Private Sub Command2_Click()

    Dim s2 As String

    hFile = CreateFile("C:\prueba.txt", GENERIC_READ, FILE_SHARE_READ, ByVal 0&, OPEN_ALWAYS, 0, 0)

    If hFile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If

    tamaño = GetFileSize(hFile, 0)

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, tamaño)

    Dim ret As Long

    'SetFilePointer hFile, 4, 0, FILE_BEGIN

    'ReadFile hFile, ByVal espacio, Len(s1) - 4, ret, ByVal 0&

    'SetFilePointer hFile, 1, 0, FILE_BEGIN

    'ReadFile hFile, ByVal espacio, Len(s1) - 1, ret, ByVal 0&

    ReadFile hFile, ByVal espacio, tamaño, ret, ByVal 0&

    s2 = String$(ret, 0)

    CopyMemory ByVal s2, ByVal espacio, ret

    Debug.Print s2

    GlobalFree espacio

    CloseHandle hFile

End Sub

Private Sub Command3_Click()

    ReDim bBytes(11)

    bBytes(1) = Asc("H"): bBytes(2) = Asc("o"): bBytes(3) = Asc("l"): bBytes(4) = Asc("a")

    bBytes(5) = Asc(" "): bBytes(6) = Asc("M"): bBytes(7) = Asc("u"): bBytes(8) = Asc("n")

    bBytes(9) = Asc("d"): bBytes(10) = Asc("o"): bBytes(11) = Asc("!")

    hFile = CreateFile("C:\otraprueba.txt", GENERIC_WRITE, FILE_SHARE_READ Or FILE_SHARE_WRITE, ByVal 0&, OPEN_ALWAYS, 0, 0)

    If hFile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If

    Dim ret As Long

    WriteFile hFile, bBytes(1), UBound(bBytes), ret, ByVal 0&

    Debug.Print ret

    CloseHandle hFile

End Sub

Private Sub Command4_Click()

    Dim s2 As String

    hFile = CreateFile("C:\otraprueba.txt", GENERIC_READ, FILE_SHARE_READ, ByVal 0&, OPEN_ALWAYS, 0, 0)

    If hFile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If

    tamaño = GetFileSize(hFile, 0)

    ReDim bBytes(tamaño)

    ReadFile hFile, bBytes(1), tamaño, ret, ByVal 0&

    For i = 1 To ret

        cadena = cadena & Chr(bBytes(i))

    Next

    Debug.Print cadena

    CloseHandle hFile

End Sub

Private Sub Command5_Click()

    hFile = CreateFile("C:\archivoslargo.txt", GENERIC_WRITE, FILE_SHARE_READ Or FILE_SHARE_WRITE, ByVal 0&, OPEN_ALWAYS, 0, 0)

    If hFile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If

    s1 = "dsfkljsdf klf sdfk jsdfk fd sdkf ds fsd fksd fsd kfsd kf " & vbNewLine & _

        "sdfjksd jsdkfj sdjf sdlkfj sd fsdkl fskdlf sdkl jfsjdk fsjdk fsdklf jsdk fdsk jfdfk" & vbNewLine & _

        "kldjfsdlkjfskdljf sdkljf sdklfj sdk jfsdkl fskld fskldfsdk fsd kfjsd kfsd klfsdf "

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1))

    CopyMemory ByVal espacio, ByVal s1, Len(s1)

    Dim ret As Long

    WriteFile hFile, ByVal espacio, Len(s1), ret, ByVal 0&

    GlobalFree espacio

    CloseHandle hFile

End Sub

Private Sub Command6_Click()

    hFile = CreateFile("C:\archivoslargo.txt", GENERIC_READ, FILE_SHARE_READ, ByVal 0&, OPEN_ALWAYS, 0, 0)

    If hFile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

        Exit Sub

    End If

    tamaño = GetFileSize(hFile, 0)

    Dim ret As Long

    Dim res As Long

    Dim salida As Boolean

    Dim s2 As String

    salida = True

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, 10)

    res = ReadFile(hFile, ByVal espacio, 10, ret, ByVal 0&)

    Do While salida

        If res And ret = 0 Then

            salida = False

        Else

            s2 = String$(ret, 0)

            CopyMemory ByVal s2, ByVal espacio, ret

            Debug.Print s2

            res = ReadFile(hFile, ByVal espacio, 10, ret, ByVal 0&)

        End If

    Loop

    GlobalFree espacio

    CloseHandle hFile

End Sub

Private Sub Command7_Click()

    Dim hOrigen As Long

    Dim hDestino As Long

    CommonDialog1.ShowOpen

    hOrigen = CreateFile(CommonDialog1.FileName, GENERIC_READ, FILE_SHARE_READ, ByVal 0&, OPEN_EXISTING, 0, 0)

    If hOrigen = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

        Exit Sub

    End If

    tamaño = GetFileSize(hOrigen, 0)

    ProgressBar1.Max = tamaño

    ProgressBar1.Min = 1

    ProgressBar1.Value = 1

    hDestino = CreateFile("C:\tArchivo." & Right(CommonDialog1.FileName, 3), GENERIC_WRITE, FILE_SHARE_READ Or FILE_SHARE_WRITE, ByVal 0&, OPEN_ALWAYS, 0, 0)

    If hDestino = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

        Exit Sub

    End If

    Dim ret As Long

    Dim ret2 As Long

    Dim res As Long

    Dim salida As Boolean

    Dim nBytes As Long

    salida = True

    nBytes = 0

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, 4096)

    res = ReadFile(hOrigen, ByVal espacio, 4096, ret, ByVal 0&)

    Do While salida

        If res And ret = 0 Then

            salida = False

        Else

            WriteFile hDestino, ByVal espacio, ret, ret2, ByVal 0&

            nBytes = nBytes + ret2

            Label1.Caption = "Bytes escritos = " & nBytes & " bytes"

            ProgressBar1.Value = nBytes

            res = ReadFile(hOrigen, ByVal espacio, 4096, ret, ByVal 0&)

            SetFilePointer hDestino, 0, ByVal 0&, FILE_END

        End If

    Loop

    GlobalFree espacio

    CloseHandle hOrigen

    CloseHandle hDestino

    Debug.Print "TERMINO!"

End Sub

Private Sub Command8_Click()

    CreateDirectory "C:\ejemplo", 0

End Sub

Private Sub Command9_Click()

    CreateDirectoryEx "C:\ejemplo", "C:\Ejemplo 2", 0

End Sub

Private Sub Form_Load()

    Command1.Caption = "1. Crear y escribir"

    Command1.Height = 600

    Command1.Left = 90

    Command1.Top = 90

    Command1.Width = 1590

    Command2.Caption = "2. Leer"

    Command2.Height = 600

    Command2.Left = 90

    Command2.Top = 765

    Command2.Width = 1590

    Command3.Caption = "3. Otra forma Crear y Escribir"

    Command3.Height = 600

    Command3.Left = 1890

    Command3.Top = 90

    Command3.Width = 1590

    Command4.Caption = "4. Otra forma Leer"

    Command4.Height = 600

    Command4.Left = 1890

    Command4.Top = 765

    Command4.Width = 1590

    Command5.Caption = "5. Archivos largos"

    Command5.Height = 600

    Command5.Left = 3690

    Command5.Top = 90

    Command5.Width = 1590

    Command6.Caption = "6. Leer archivos largos"

    Command6.Height = 600

    Command6.Left = 3690

    Command6.Top = 765

    Command6.Width = 1590

    Command7.Caption = "7. Transferencia de informacion"

    Command7.Height = 600

    Command7.Left = 5490

    Command7.Top = 90

    Command7.Width = 1590

    Command8.Caption = "8. Crear Directorio"

    Command8.Height = 600

    Command8.Left = 5490

    Command8.Top = 765

    Command8.Width = 1590

    Command9.Caption = "9. Crear Directorio Ex"

    Command9.Height = 600

    Command9.Left = 7290

    Command9.Top = 90

    Command9.Width = 1590

    Command10.Caption = "10. Borrar Directorio"

    Command10.Height = 600

    Command10.Left = 7290

    Command10.Top = 765

    Command10.Width = 1590

    Label1.Caption = vbNullString

    Label1.Height = 465

    Label1.Left = 135

    Label1.Top = 1530

    Label1.Width = 6765

    ProgressBar1.Height = 285

    ProgressBar1.Left = 135

    ProgressBar1.Top = 1755

    ProgressBar1.Width = 6810

    Me.Width = 9390

    Me.Height = 2610

End Sub
Antes de correrlo vamos a dejar en claro ciertas cositas, en primer lugar se habrán dado cuenta de los comentarios colocados en la declaración de algunas funciones, ejemplo:

'Private Declare Function CreateFile Lib "kernel32" Alias "CreateFileA" (ByVal lpFileName As String, ByVal dwDesiredAccess As Long, ByVal dwShareMode As Long, lpSecurityAttributes As SECURITY_ATTRIBUTES, ByVal dwCreationDisposition As Long, ByVal dwFlagsAndAttributes As Long, ByVal hTemplateFile As Long) As Long

Private Declare Function CreateFile Lib "kernel32" Alias "CreateFileA" (ByVal lpFileName As String, ByVal dwDesiredAccess As Long, ByVal dwShareMode As Long, lpSecurityAttributes As Long, ByVal dwCreationDisposition As Long, ByVal dwFlagsAndAttributes As Long, ByVal hTemplateFile As Long) As Long

Su única diferencia es lo que esta marcado en azul. La estructura SECURITY_ATTRIBUTES es una estructura para NT, que no vamos a usar, por ende para que no sea un fastido, declarar dicha estructura, cambio la función quitando en la declaración el parámetro SECURITY_ATTRIBUTES y coloco Long, con la finalidad de poder pasarle 0, cuando utilice la función.

Como pueden ver hacemos acciones simples, pero que en código no son nada simple, dicha explicación la haré botón por botón, pero en este momento describiré 3 funciones claves, que vamos a ver en prácticamente todos los botones.

CreateFile

Api: Declare Function CreateFile Lib "kernel32" Alias "CreateFileA" (ByVal lpFileName As String, ByVal dwDesiredAccess As Long, ByVal dwShareMode As Long, lpSecurityAttributes As Long, ByVal dwCreationDisposition As Long, ByVal dwFlagsAndAttributes As Long, ByVal hTemplateFile As Long) As Long

Función encargada de crear o abrir un archivo. 

LpFileName = Como primer parámetro tenemos el nombre del archivo que queremos crear o abrir.

DwDesiredAccess = Especifica el tipo de acceso, esto parámetro puede tomar alguna o la combinacion de estas dos variables:

Const GENERIC_READ = &H80000000 ‘ Se especifica que el archivo será abierto o creado para solo ser leído.

Const GENERIC_WRITE = &H40000000 ‘ Se especifica que el archivo será abierto o creado para solo escritura.

Pueden usar OR para combinar ambas constantes.

DwShareMode = con este parámetro especificamos como va ser compartido el archivo creado o abierto. Puede tomar la combinación de los siguientes valores:

Const FILE_SHARE_READ = &H1 
Const FILE_SHARE_WRITE = &H2

Ambos parámetro son utilizados para especificar que se comparte el archivo solo para lectura, escritura o ambos. Recuerden que un archivo puede ser usado por varios programas al mismo tiempo.

LpSecurityAttributes = Aquí es donde va la estructura SECURITY_ATTRIBUTES  pero en nuestro caso no será utilizada y colocaremos 0. Aquellos que usan NT y están interesados pueden consultar la documentación en MSDN.

DwCreationDisposition = Con este parámetro se especifica la acciones a tomar, en caso de que exista el archivo o no exista. Para verlo mejor veamos los valores que puede tomar:

Const CREATE_NEW = 1 ‘ Crea el archivo, en caso de existir el archivo la función retorna un error es decir falla la ejecución de la función.

Const CREATE_ALWAYS = 2 ‘ Crea el archivo, en caso de existir el archivo, este es sobrescrito, es decir, se pierde la información existente.

Const OPEN_EXISTING = 3 ‘  Sirve para solo abrir el archivo cuando ya sabemos que existe, en caso de no existir la función retorna un error.

Const OPEN_ALWAYS = 4 ‘ Abre el archivo ya existente, en caso de no existir la función lo crea si el valor CREATE_NEW esta especificado.

Const TRUNCATE_EXISTING = 5 ‘ Abre el archivo existente, y una vez abierto este toma el tamaño de 0 bytes. La función falla si el archivo no existe.

DwFlagsAndAttributes = Con este parámetro especificamos los atributos que va a poseer dicho archivo, puede tomar un valor o la combinacion de los siguientes valores:

Const FILE_ATTRIBUTE_ARCHIVE = &H20 ‘ Con este especificamos que lo que estamos creando es un archivo.

Const FILE_ATTRIBUTE_HIDDEN = &H2 ‘ Con este especificamos que el archivo sea oculto.

Const FILE_ATTRIBUTE_NORMAL = &H80 ‘ El archivo no posee de ningún atributo en particular.

Const FILE_ATTRIBUTE_OFFLINE = &H1000 ‘ Con este se especifica que la información del archivo no esta habilitada inmediatamente.

Const FILE_ATTRIBUTE_READONLY = &H1 ‘  El archivo es de solo lectura.

Const FILE_ATTRIBUTE_SYSTEM = &H4 ‘ El archivo forma parte del sistema, es decir, un archivo que el sistema operativo usa.

Const FILE_ATTRIBUTE_TEMPORARY = &H100 ‘ El archivo va ser  utilizado de manera temporal. 

Con estos parámetros se pueden usar también los siguientes, que solo son nombrados ya que no son usado en este manual:

Private Const FILE_FLAG_WRITE_THROUGH = &H80000000

Private Const FILE_FLAG_OVERLAPPED = &H40000000

Private Const FILE_FLAG_NO_BUFFERING = &H20000000

Private Const FILE_FLAG_RANDOM_ACCESS = &H10000000

Private Const FILE_FLAG_SEQUENTIAL_SCAN = &H8000000

Private Const FILE_FLAG_DELETE_ON_CLOSE = &H4000000

Private Const FILE_FLAG_BACKUP_SEMANTICS = &H2000000

Private Const FILE_FLAG_POSIX_SEMANTICS = &H1000000

Private Const FILE_FLAG_OPEN_REPARSE_POINT = &H200000

Private Const FILE_FLAG_OPEN_NO_RECALL = &H100000
HTemplateFile = en este parámetro se le pasa el “handle” de un archivo abierto con anterioridad, ¿Para que sirve eso?. Lo que se hace con esto es heredar o tomar los parámetros de ese archivo. Es decir, si el archivo abierto tiene el atributo HIDDEN, el archivo que vayamos a crear toma también dicho atributo. Lo veremos mas claro cuando vayamos a crear directorios.

Estos son todos los parámetros, si la función se ejecuta con éxito dicha función retorna el “handle” del objeto abierto, que en este caso es un archivo. En caso de fallar este retorna:

Const INVALID_HANDLE_VALUE = -1

Para un error mas especifico usamos la función GetLastError, función que veremos mas adelante.

Ahora voy a mostrar las dos funciones que utilizamos para leer el archivo y escribir en él, algo que tiene que tener en cuenta aparte de haber entendido el Capítulo  de Memoria, es que un archivo es una arreglo de bytes. Es importante que tengan esto en mente ya que lo necesitaremos al momento de explicar el código.

WriteFile

Api: Declare Function WriteFile Lib "kernel32" (ByVal hFile As Long, lpBuffer As Any, ByVal nNumberOfBytesToWrite As Long, lpNumberOfBytesWritten As Long, lpOverlapped As Long) As Long

Api: Declare Function WriteFile Lib "kernel32" (ByVal hFile As Long, lpBuffer As Any, ByVal nNumberOfBytesToWrite As Long, lpNumberOfBytesWritten As Long, lpOverlapped As OVERLAPPED) As Long

Aquí pongo dos declaraciones ya que dense cuenta que el ultimo parámetro no es igual, y esto se debe a que no vamos a usar la estructura OVERLAPPED, ¿Por qué?, para ser sincero, es por que yo no le he visto uso a este parámetro, OJO, no estoy diciendo que no sirve para nada esta estructura, solo que YO no he visto la necesidad de usarla para hacer mis proyectos, por lo que no me he preocupado de estudiarla.

Esta función como obviamente se imaginaran sirve para escribir en un archivo.

HFile = como primer parámetro le pasamos el “handle” que nos regresa CreateFile, del archivo que se encuentra abierto, para que la función sirva, el archivo tuvo que ser creado o abierto especificando la constante GENERIC_WRITE.
LpBuffer = puntero de memoria en donde se encuentra la información  que queremos escribir en el archivo.

NNumberOfBytesToWrite = numero de bytes que vamos a escribir.

LpNumberOfBytesWritten = variable que después de haber llamado a la función contendrá la cantidad de bytes que fueron escritos.

LpOverlapped  = parámetro en donde se le pasa la estructura OVERLAPPED pero que en nuestro caso será 0.

Si la función se ejecuta con éxito retorna un valor distinto de 0, en caso de retornar 0 implica que hubo alguna falla en su ejecución.

ReadFile

Api: Declare Function ReadFile Lib "kernel32" (ByVal hFile As Long, lpBuffer As Any, ByVal nNumberOfBytesToRead As Long, lpNumberOfBytesRead As Long, lpOverlapped As OVERLAPPED) As Long

Api: Declare Function ReadFile Lib "kernel32" (ByVal hFile As Long, lpBuffer As Any, ByVal nNumberOfBytesToRead As Long, lpNumberOfBytesRead As Long, lpOverlapped As Long) As Long

Función encargada de leer la información contenida en un archivo.

HFile = como primer parámetro le pasamos el “handle” que nos regresa CreateFile, del archivo que se encuentra abierto, para que la función sirva, el archivo tuvo que ser creado o abierto especificando la constante GENERIC_READ.
LpBuffer = puntero de memoria en donde se almacenará la información. 

NNumberOfBytesToWrite = numero de bytes que vamos a leer.

LpNumberOfBytesWritten = variable que después de haber llamado a la función contendrá la cantidad de bytes que fueron leídos.

LpOverlapped  = parámetro en donde se le pasa la estructura OVERLAPPED pero que en nuestro caso será 0.

Ahora que ya hemos visto las funciones claves, para entender el funcionamiento de cada una vamos a los ejemplos, y las funciones que aparezcan nuevas serán explicadas a medida que expliquemos el funcionamiento del programa.

Command1

hFile = CreateFile("C:\prueba.txt", GENERIC_WRITE, FILE_SHARE_READ Or FILE_SHARE_WRITE, ByVal 0&, OPEN_ALWAYS, 0, 0) ‘ Creo el archivo especificando GENERIC_WRITE, donde doy a entender que vamos a escribir en él.

      .   

      .    

    s1 = "Hola Mundo!"

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1)) ‘ Creo un espacio en memoria que contendrá la información de la variable “s1”, doy por entendido que leyeron el Capítulo  de Memoria.

    CopyMemory ByVal espacio, ByVal s1, Len(s1)

    Dim ret As Long

    WriteFile hFile, ByVal espacio, Len(s1), ret, ByVal 0& ‘ Escribo la información que apunta la variable “espacio”, en el archivo abierto “hFile”, el valor de “ret” después de ejecutar la función es igual a Len(s1) ya que escribimos toda la cadena

    GlobalFree espacio ‘  Liberamos el espacio de memoria que usamos.

Aquí hago un STOP, para explicar la siguiente función API y luego continuo con el código, aunque en este momento en C:\ tienen un archivo “prueba.txt” que contiene en su interior “Hola Mundo!”

SetFilePointer

Api: Declare Function SetFilePointer Lib "kernel32" (ByVal hFile As Long, ByVal lDistanceToMove As Long, lpDistanceToMoveHigh As Long, ByVal dwMoveMethod As Long) As Long

Función que sirve para mover el puntero del archivo ¿Puntero?. En un archivo existe un puntero dicho puntero es utilizado para saber en donde estamos en el archivo, es similar al puntero de bases de datos, con la diferencia que aquí hablamos de cualquier tipo de archivo.

HFile = como primer parámetro le pasamos el “handle” que nos regresa CreateFile, del archivo que se encuentra abierto.

LDistanceToMove, lpDistanceToMoveHigh = estas dos variables forman un valor de 64 Bits, que al combinarse especifican la cantidad de bytes que se va a mover el puntero del archivo. La primera especifica la parte baja de 32 bits y la segunda la parte alta. Particularmente en los ejemplos que use, utilice LdistanceToMove, para especificar la cantidad de bytes que quería mover colocando 0 en lpDistanceToMoveHigh. Eso lo veremos mas adelante.

DwMoveMethod = Este parámetro puede tomar los siguientes valores:

Const FILE_BEGIN = 0 ‘ Coloca el puntero en el inicio del archivo

Const FILE_END = 2 ‘ Coloca el puntero al final del archivo.

Ya explicada dicha función continuemos con el código:

    'SetFilePointer hFile, 0, ByVal 0&, FILE_BEGIN

    SetFilePointer hFile, 0, ByVal 0&, FILE_END ‘ Aquí lo que hago es llevar el puntero al final del archivo para continuar escribiendo.

    s1 = "como estas"

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1))

    CopyMemory ByVal espacio, ByVal s1, Len(s1)

    WriteFile hFile, ByVal espacio, Len(s1), ret, ByVal 0&

    GlobalFree espacio

    CloseHandle hFile

Aplico el mismo procedimiento de escritura explicado arriba. Pero quiten el comentario de la línea:

'SetFilePointer hFile, 0, ByVal 0&, FILE_BEGIN.

Y pónganle comentario a SetFilePointer hFile, 0, ByVal 0&, FILE_END, luego borren el archivo creado, ejecuten el programa y vean el archivo y dense cuenta de la diferencia.

Command2

Dim s2 As String

    hFile = CreateFile("C:\prueba.txt", GENERIC_READ, FILE_SHARE_READ, ByVal 0&, OPEN_ALWAYS, 0, 0) ‘ Abro el archivo de modo LECTURA.

    If hFile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If

    tamaño = GetFileSize(hFile, 0)  ‘ Obtengo su tamaño.

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, tamaño) ‘ Reservo la memoria donde almacenare la información.

    Dim ret As Long

    'SetFilePointer hFile, 4, 0, FILE_BEGIN

    'ReadFile hFile, ByVal espacio, Len(s1) - 4, ret, ByVal 0&

    'SetFilePointer hFile, 1, 0, FILE_BEGIN

    'ReadFile hFile, ByVal espacio, Len(s1) - 1, ret, ByVal 0&

    ReadFile hFile, ByVal espacio, tamaño, ret, ByVal 0& ‘ Recupero la información cuyo tamaño en bytes será lo que contenga la variable “tamaño” y lo almaceno en la dirección que apunta “espacio”.

    s2 = String$(ret, 0)

    CopyMemory ByVal s2, ByVal espacio, ret

    Debug.Print s2 ‘Muestro el contenido del archivo.

    GlobalFree espacio

    CloseHandle hFile
Si aprietan el botón 1, en donde escribimos y luego el 2, en la ventana “Inmediato” de VB verán el contenido del archivo. Que sucede con estas líneas de código:

    'SetFilePointer hFile, 4, 0, FILE_BEGIN

    'ReadFile hFile, ByVal espacio, Len(s1) - 4, ret, ByVal 0&

    'SetFilePointer hFile, 1, 0, FILE_BEGIN

    'ReadFile hFile, ByVal espacio, Len(s1) - 1, ret, ByVal 0&

Si ponen en comentario a esta línea:

    ReadFile hFile, ByVal espacio, tamaño, ret, ByVal 0&

Y le quitan el comentario a las líneas marcadas de azul, se darán cuenta que la información que retorna no es la misma, es mas pequeño ya que moví el puntero del archivo 4 bytes, es decir, si en el archivo estaba la cadena : “Hola Mundo!”, pues lo que obtendremos será “Mundo!”, ya que movimos el puntero 4 bytes hacia la derecha. Igual sucede con las líneas de color rojo.

Command3

Ya sabemos que en un archivo lo que pasamos es una serie de bytes, que representan alguna información, hemos utilizado la variable “espacio” como puntero de dicha serie de bytes que queremos almacenar, pero el command3 muestra otra forma de almacenar información, pero a través de arreglo, dicho arreglo deberá de ser de tipo byte, y cada casilla tendrá que tener el valor que queremos almacenar.

    ReDim bBytes(11) ‘ Declaro un arreglo de 11 casillas del tipo byte.

    bBytes(1) = Asc("H"): bBytes(2) = Asc("o"): bBytes(3) = Asc("l"): bBytes(4) = Asc("a")

    bBytes(5) = Asc(" "): bBytes(6) = Asc("M"): bBytes(7) = Asc("u"): bBytes(8) = Asc("n")

    bBytes(9) = Asc("d"): bBytes(10) = Asc("o"): bBytes(11) = Asc("!") ‘ Almaceno en cada casilla el ascii correspondiente a cada letra del contenido que vamos a guardar.

    hFile = CreateFile("C:\otraprueba.txt", GENERIC_WRITE, FILE_SHARE_READ Or FILE_SHARE_WRITE, ByVal 0&, OPEN_ALWAYS, 0, 0) ‘ Creo el archivo

    .

    .

    Dim ret As Long

    WriteFile hFile, bBytes(1), UBound(bBytes), ret, ByVal 0& ‘ Paso el arreglo que contiene la información a guardar.

    .

    .
Vean que se obtiene exactamente lo mismo. Una pregunta podría ser ¿Por qué “espacio” lo pasas con ByVal, y el arreglo sin Byval? Buena pregunta. Ese parámetro de la función WriteFile, (y estoy hablando de “LpBuffer”) es un parámetro por pase por referencia. La variable “espacio” NO, y ojo NO contiene el contenido de la cadena, es decir, “espacio” no tiene almacenado “Hola Mundo!”, lo que tiene “espacio” almacenado es la dirección en memoria en donde se encuentra dicha cadena. Si yo quito Byval, pues entonces la función tomaría la dirección como valor y NO la cadena, es decir, quedaría almacenado el contenido de la variable “espacio” que no es mas que un simple numero, y claro, daría error la función ya que el tamaño a guardar es mayor que el contenido de la variable “espacio”.

En cambio bBytes, es un arreglo, y todos sabemos que la primera posición, apunta a la primera posición en la memoria en donde se encuentra los bytes que queremos almacenar, no uso ByVal ya que aquí si me interesa que tome el pase por referencia de la primera posición. Se que es confuso y mas para aquellos que no pueden tener experiencia en C, pero siéntense analizar un poco y se darán cuenta lo que sucede.

Command4

Command4 es el equivalente al Command2, pero usando arreglo:

NOTA: omito las líneas que carecen de importancia.

    hFile = CreateFile("C:\otraprueba.txt", GENERIC_READ, FILE_SHARE_READ, ByVal 0&, OPEN_ALWAYS, 0, 0)  ‘ Abro el archivo creado con Command3

    tamaño = GetFileSize(hFile, 0) ‘  Obtengo el tamaño

    ReDim bBytes(tamaño) ‘ Redimensiono el arreglo para que soporte toda la información

    ReadFile hFile, bBytes(1), tamaño, ret, ByVal 0& ‘ Luego leo el contenido. Que va empezar almacenándose en la posición 1 del arreglo.

    For i = 1 To ret

        cadena = cadena & Chr(bBytes(i))

    Next ’ Con este FOR construyo la cadena resultante.

Hay una función que hemos usado y que de seguro han entendido lo que hace para vamos a mostrarla con detalle:

GetFileSize

Api: Declare Function GetFileSize Lib "kernel32" (ByVal hFile As Long, lpFileSizeHigh As Long) As Long

función que se encarga de tomar el tamaño en bytes de un archivo, como primer parámetro le pasamos el “handle” del archivo y como segundo parámetro, en verdad no le he visto utilidad por lo que para nuestro propósito le pasamos 0. 

La función lo que retorna es el tamaño en bytes del archivo.

Command5

Este botón hace exactamente lo mismo que command1, con la diferencia que almacena un valor de tipo String mas grande. Esto sirve para explicar el botón que sigue en donde veremos como hacer un recorrido del archivo.

Command6
    salida = True ‘ Variable de control que sirve para controlar cuando el puntero llegue a fin de archivo.

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, 10)

    res = ReadFile(hFile, ByVal espacio, 10, ret, ByVal 0&) ‘ Vamos a leer el archivo de 10 en 10, es decir, tomamos primero 10 bytes y así sucesivamente hasta llegar al EOF o fin de archivo.

    Do While salida

        If res And ret = 0 Then ‘ Este es la condición de EOF, es decir, cuando res = true y ret = 0 esto implica que la función ha llegado al final del archivo por ende no hay mas nada que leer, por lo que cambio salida = false para salir del bucle.

            salida = False

        Else ‘ En caso de no ser fin de archivo

            s2 = String$(ret, 0) 

            CopyMemory ByVal s2, ByVal espacio, ret ‘ Copio los primeros 10 bytes tomados en la variable s2.

            Debug.Print s2 ‘ Voy imprimiendo en la ventana “inmediato” los bytes que voy recuperando.

            res = ReadFile(hFile, ByVal espacio, 10, ret, ByVal 0&) ‘ Procedo a leer los 10 bytes siguientes. Almacenándolos en el mismo espacio de memoria que reserve al inicio.

        End If

    Loop

Command7
El command7 es un botón muy interesante ya que vamos hacer una copia EXACTA de un archivo sin usar ninguna función de copia, sino puro WriteFile y ReadFile.

Voy a saltarme las primeras líneas en donde no deben de tener problema de entender, así que mostrare las mas importantes:

    salida = True ‘ Variable de control del bucle WHILE

    nBytes = 0

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, 4096) ‘ Reserva 4K, lo que significa que vamos a ir leyendo el archivo de 4K en 4K.

    res = ReadFile(hOrigen, ByVal espacio, 4096, ret, ByVal 0&) ‘ Leo los primeros 4K.

    Do While salida

        If res And ret = 0 Then ‘ Si es fin de archivo.

            salida = False ‘ Cambio el valor de la variable “salida” para salir del bucle.

        Else ‘ Si no es fin de archivo.

            WriteFile hDestino, ByVal espacio, ret, ret2, ByVal 0& ‘ Escribo los primeros 4K en el archivo “hDestino”.

            nBytes = nBytes + ret2 ‘ Llevo la cuenta de la cantidad de bytes leído.

            Label1.Caption = "Bytes escritos = " & nBytes & " bytes" ‘ Muestro los bytes leídos.

            ProgressBar1.Value = nBytes 

            res = ReadFile(hOrigen, ByVal espacio, 4096, ret, ByVal 0&) ‘ Leo los próximos 4K del archivo “origen”

            SetFilePointer hDestino, 0, ByVal 0&, FILE_END ‘ Me aseguro que el puntero del archivo “hDestino” este en el final, para ir almacenando los bytes uno detrás de otro.

        End If

    Loop

Y listo, si quieren probarlo agarren un archivo grafico, y vean que la copia, tiene la misma imagen que el archivo original.

Command8
botón utilizado para crear un directorio.

CreateDirectory

Api: Declare Function CreateDirectory Lib "kernel32" Alias "CreateDirectoryA" (ByVal lpPathName As String, lpSecurityAttributes As Long) As Long

Api: Declare Function CreateDirectory Lib "kernel32" Alias "CreateDirectoryA" (ByVal lpPathName As String, lpSecurityAttributes As SECURITY_ATTRIBUTES) As Long

Esta función es utilizada para crear un nuevo directorio, como primer parámetro le pasamos el nombre y ubicación del directorio, y como segundo parámetro en la función original debería de pasársele la estructura SECURITY_ATTRIBUTES, pero como no la vamos usar, usamos la función que posee dicho parámetro declarado de tipo long, y pasamos 0, cuando ejecutemos la función.

Command9
botón utilizado para crear un directorio que toma los atributos de otro directorio.

CreateDirectoryEx

Api:Declare Function CreateDirectoryEx Lib "kernel32" Alias "CreateDirectoryExA" (ByVal lpTemplateDirectory As String, ByVal lpNewDirectory As String, lpSecurityAttributes As Long) As Long

Api: Declare Function CreateDirectoryEx Lib "kernel32" Alias "CreateDirectoryExA" (ByVal lpTemplateDirectory As String, ByVal lpNewDirectory As String, lpSecurityAttributes As SECURITY_ATTRIBUTES) As Long

Función que crea un directorio usando los atributos de un directorio que nos sirve como plantilla. Es decir, el directorio creado con el Command8, asígnenle la propiedad de oculto, y luego presionen el botón Command9, y se darán cuenta que el nuevo directorio “C:\Ejemplo 2”, se creo y también es una carpeta oculta ya que heredo los atributos del directorio “C:\Ejemplo”.

Como primer parámetro le pasamos el directorio que nos sirve como plantilla, como segundo parámetro le pasamos el directorio que queremos crear y como tercer parámetro pasamos 0.

Command10
botón utilizado para eliminar a un directorio.

RemoveDirectory

Api: Declare Function RemoveDirectory Lib "kernel32" Alias "RemoveDirectoryA" (ByVal lpPathName As String) As Long

función utilizada para eliminar un directorio del disco, si la eliminación es un éxito dicha función retorna un valor distinto de 0, en caso de fracasar retornara 0. Y como único parámetro le pasamos el directorio que queremos eliminar.

Se habrán dado cuenta que no es nada fácil trabajar con archivos usando las librerías de Windows, pero algunas veces dichas librerías nos proporcionan de cierto control y versatilidad que no encontraremos en los controles que se encargan de la realización de estas funciones.

BUSCANDO ARCHIVOS

Ahora vamos a ver un proyecto bien curioso y muy interesante ya que vamos a aprender a como buscar un archivo en un directorio. Y veremos las funciones involucradas en dicho proceso. Para este proyecto son necesarios los siguientes controles: 4 Botones “command”, 2 etiquetas o “label”, 2 Listbox, 3 cajas de texto o “TextBox”. Luego pegan este código:

Private Const STATUS_WAIT_0 = (&H0)

Private Const STATUS_ABANDONED_WAIT_0 = (&H80)

Private Const WAIT_ABANDONED = ((STATUS_ABANDONED_WAIT_0) + 0)

Private Const WAIT_OBJECT_0 = ((STATUS_WAIT_0) + 0)

Private Const WAIT_TIMEOUT = 258&

Private Const FILE_NOTIFY_CHANGE_FILE_NAME = &H1

Private Const FILE_NOTIFY_CHANGE_ATTRIBUTES = &H4

Private Const FILE_NOTIFY_CHANGE_DIR_NAME = &H2

Private Const FILE_NOTIFY_CHANGE_SIZE = &H8

Private Const FILE_NOTIFY_CHANGE_LAST_WRITE = &H10

Private Const FILE_NOTIFY_CHANGE_SECURITY = &H100

Private Const FILE_ATTRIBUTE_ARCHIVE = &H20

Private Const FILE_ATTRIBUTE_COMPRESSED = &H800

Private Const FILE_ATTRIBUTE_DIRECTORY = &H10

Private Const FILE_ATTRIBUTE_ENCRYPTED = &H4000

Private Const FILE_ATTRIBUTE_HIDDEN = &H2

Private Const FILE_ATTRIBUTE_NORMAL = &H80

Private Const FILE_ATTRIBUTE_OFFLINE = &H1000

Private Const FILE_ATTRIBUTE_READONLY = &H1

Private Const FILE_ATTRIBUTE_REPARSE_POINT = &H400

Private Const FILE_ATTRIBUTE_SPARSE_FILE = &H200

Private Const FILE_ATTRIBUTE_SYSTEM = &H4

Private Const FILE_ATTRIBUTE_TEMPORARY = &H100

Private Const ERROR_NO_MORE_FILES = 18&

Private Const MAX_PATH = 260

Private Const OFS_MAXPATHNAME = 128

Private Const MAXWORD = &HFFFF

Private Const INVALID_HANDLE_VALUE = -1

Private Type FILETIME

    dwLowDateTime As Long

    dwHighDateTime As Long

End Type

Private Type WIN32_FIND_DATA

    dwFileAttributes As Long

    ftCreationTime As FILETIME

    ftLastAccessTime As FILETIME

    ftLastWriteTime As FILETIME

    nFileSizeHigh As Long

    nFileSizeLow As Long

    dwReserved0 As Long

    dwReserved1 As Long

    cFileName As String * MAX_PATH

    cAlternate As String * 14

End Type

Private Type SYSTEMTIME

    wYear As Integer

    wMonth As Integer

    wDayOfWeek As Integer

    wDay As Integer

    wHour As Integer

    wMinute As Integer

    wSecond As Integer

    wMilliseconds As Integer

End Type

Private Declare Function FindClose Lib "kernel32" (ByVal hFindFile As Long) As Long

Private Declare Function FindCloseChangeNotification Lib "kernel32" (ByVal hChangeHandle As Long) As Long

Private Declare Function FindFirstChangeNotification Lib "kernel32" Alias "FindFirstChangeNotificationA" (ByVal lpPathName As String, ByVal bWatchSubtree As Long, ByVal dwNotifyFilter As Long) As Long

Private Declare Function FindFirstFile Lib "kernel32" Alias "FindFirstFileA" (ByVal lpFileName As String, lpFindFileData As WIN32_FIND_DATA) As Long

Private Declare Function FindNextChangeNotification Lib "kernel32" (ByVal hChangeHandle As Long) As Long

Private Declare Function FindNextFile Lib "kernel32" Alias "FindNextFileA" (ByVal hFindFile As Long, lpFindFileData As WIN32_FIND_DATA) As Long

Private Declare Function FileTimeToLocalFileTime Lib "kernel32" (lpFileTime As FILETIME, lpLocalFileTime As FILETIME) As Long

Private Declare Function FileTimeToSystemTime Lib "kernel32" (lpFileTime As FILETIME, lpSystemTime As SYSTEMTIME) As Long

Private Declare Function GetCurrentDirectory Lib "kernel32" Alias "GetCurrentDirectoryA" (ByVal nBufferLength As Long, ByVal lpBuffer As String) As Long

Private Declare Function SetCurrentDirectory Lib "kernel32" Alias "SetCurrentDirectoryA" (ByVal lpPathName As String) As Long

Private Declare Function GetLastError Lib "kernel32" () As Long

Private Declare Function WaitForSingleObject Lib "kernel32" (ByVal hHandle As Long, ByVal dwMilliseconds As Long) As Long

Dim ftime As FILETIME, ftime2 As FILETIME

Dim stime As SYSTEMTIME

Dim wData As WIN32_FIND_DATA

Dim cDirectory As String

Dim hBusqueda As Long

Dim hChequeo As Long

Private Sub Command1_Click()

    Dim encontrado As Boolean

    encontrado = True

    hBusqueda = FindFirstFile(Text1.Text, wData)

    If hBusqueda = INVALID_HANDLE_VALUE Then

        MsgBox "Archivo no encontrado"

        Exit Sub

    End If

    List1.AddItem wData.cFileName

    If wData.dwFileAttributes And FILE_ATTRIBUTE_ARCHIVE Then

        List1.AddItem "Este es un archivo o directorio"

    Else

        List1.AddItem "Este no es un archivo o directorio"

    End If

    If wData.dwFileAttributes And FILE_ATTRIBUTE_COMPRESSED Then

        List1.AddItem "Este es un archivo o directorio comprimido"

    Else

        List1.AddItem "Este no es un archivo o directorio comprimido"

    End If

    If wData.dwFileAttributes And FILE_ATTRIBUTE_DIRECTORY Then

        List1.AddItem "Este es un directorio "

    Else

        List1.AddItem "Este no es un directorio"

    End If

    If wData.dwFileAttributes And FILE_ATTRIBUTE_ENCRYPTED Then

        List1.AddItem "Este es un archivo encriptado "

    Else

        List1.AddItem "Este no es un archivo encriptado"

    End If

    If wData.dwFileAttributes And FILE_ATTRIBUTE_HIDDEN Then

        List1.AddItem "Este es un archivo o directorio oculto "

    Else

        List1.AddItem "Este no es un archivo o directorio oculto"

    End If

    If wData.dwFileAttributes And FILE_ATTRIBUTE_NORMAL Then

        List1.AddItem "Este es un archivo o directorio que no tiene atributos "

    Else

        List1.AddItem "Este es un archivo o directorio que tiene atributo"

    End If

    If wData.dwFileAttributes And FILE_ATTRIBUTE_OFFLINE Then

        List1.AddItem "Este es un archivo o directorio que no puede ser usado en este momento "

    Else

        List1.AddItem "Este es un archivo o directorio que puede ser usado en este momento"

    End If

    If wData.dwFileAttributes And FILE_ATTRIBUTE_READONLY Then

        List1.AddItem "Este es un archivo o directorio de solo lectura "

    Else

        List1.AddItem "Este es un archivo o directorio que no es de solo lectura"

    End If

    If wData.dwFileAttributes And FILE_ATTRIBUTE_SYSTEM Then

        List1.AddItem "Este es un archivo o directorio usado por el sistema "

    Else

        List1.AddItem "Este es un archivo o directorio que no es del sistema"

    End If

    FileTimeToLocalFileTime wData.ftCreationTime, ftime

    FileTimeToSystemTime ftime, stime

    List1.AddItem " Dia = " & stime.wDay

    List1.AddItem " Dia de la semana = " & stime.wDayOfWeek

    List1.AddItem " Hora = " & stime.wHour

    List1.AddItem " Minuto = " & stime.wMinute

    List1.AddItem " Segundos = " & stime.wSecond

    List1.AddItem " Milisegundos = " & stime.wMilliseconds

    List1.AddItem " MEs = " & stime.wMonth

    List1.AddItem " Año = " & stime.wYear

    FileTimeToLocalFileTime wData.ftLastAccessTime, ftime

    FileTimeToSystemTime ftime, stime

    List1.AddItem " Ultimo acceso Dia = " & stime.wDay

    List1.AddItem " Ultimo acceso Dia de la semana = " & stime.wDayOfWeek

    List1.AddItem " Ultimo acceso Hora = " & stime.wHour

    List1.AddItem " Ultimo acceso Minuto = " & stime.wMinute

    List1.AddItem " Ultimo acceso Segundos = " & stime.wSecond

    List1.AddItem " Ultimo acceso MEs = " & stime.wMonth

    List1.AddItem " Ultimo acceso Año = " & stime.wYear

    FileTimeToLocalFileTime wData.ftCreationTime, ftime

    FileTimeToSystemTime ftime, stime

    List1.AddItem " Ultima escritura Dia = " & stime.wDay

    List1.AddItem " Ultima escritura Dia de la semana = " & stime.wDayOfWeek

    List1.AddItem " Ultima escritura Hora = " & stime.wHour

    List1.AddItem " Ultima escritura Minuto = " & stime.wMinute

    List1.AddItem " Ultima escritura Segundos = " & stime.wSecond

    List1.AddItem " Ultima escritura MEs = " & stime.wMonth

    List1.AddItem " Ultima escritura Año = " & stime.wYear

    List1.AddItem "Tamaño = " & ((wData.nFileSizeHigh * maxdword) + wData.nFileSizeLow)

    FindClose hBusqueda

End Sub

Private Sub Command2_Click()

    SetCurrentDirectory Text2.Text

    Dim res As Long

    cDirectory = Space(MAX_PATH)

    res = GetCurrentDirectory(MAX_PATH, cDirectory)

    Label1 = Left(cDirectory, res)

End Sub

Private Sub Command3_Click()

    Dim encontrado As Boolean

    encontrado = True

    List2.Clear

    hBusqueda = FindFirstFile(Text3.Text, wData)

    If hBusqueda = INVALID_HANDLE_VALUE Then

        MsgBox "Archivo no encontrado"

        Exit Sub

    End If

    Do While encontrado

        List2.AddItem wData.cFileName

        If Not FindNextFile(hBusqueda, wData) Then

            If GetLastError = ERROR_NO_MORE_FILES Then

                encontrado = False

            End If

        End If

    Loop

    FindClose hBusqueda

End Sub

Private Sub Command4_Click()

    hChequeo = FindFirstChangeNotification("C:\", False, FILE_NOTIFY_CHANGE_FILE_NAME)

    Dim hObject As Long

    Dim bucle As Boolean

    bucle = True

    hObject = WaitForSingleObject(hChequeo, 50000)

    Select Case hObject

    Case WAIT_OBJECT_0

        Label2.Caption = "EPA POR QUE ANDAS CREANDO ARCHIVOS"

    Case WAIT_TIMEOUT

        MsgBox "TIEMPO EXCEDIDO"

    End Select

    FindCloseChangeNotification hChequeo

End Sub

Private Sub Form_Activate()

    Dim res As Long

    cDirectory = Space(MAX_PATH)

    res = GetCurrentDirectory(MAX_PATH, cDirectory)

    Label1 = Left(cDirectory, res)

End Sub

Private Sub Form_Load()

    Command1.Caption = "Buscar Archivo Individual"

    Command1.Height = 465

    Command1.Left = 90

    Command1.Top = 495

    Command1.Width = 1905

    Command2.Caption = "Cambiar a directorio actual"

    Command2.Height = 465

    Command2.Left = 90

    Command2.Top = 1395

    Command2.Width = 1905

    Command3.Caption = "Buscar varios archivos"

    Command3.Height = 464

    Command3.Left = 4095

    Command3.Top = 495

    Command3.Width = 1905

    Command4.Caption = "Chequeo"

    Command4.Height = 465

    Command4.Left = 8550

    Command4.Top = 90

    Command4.Width = 2670

    Label1.Caption = vbNullString

    Label1.Height = 510

    Label1.Left = 90

    Label1.Top = 1935

    Label1.Width = 3795

    Label2.Caption = vbNullString

    Label2.Height = 1050

    Label2.Left = 8550

    Label2.Top = 675

    Label2.Width = 2670

    List1.Height = 2205

    List1.Left = 90

    List1.Top = 2565

    List1.Width = 3795

    List2.Height = 2205

    List2.Left = 4095

    List2.Top = 2565

    List2.Width = 3795

    Text1.Text = vbnulltring

    Text1.Height = 285

    Text1.Left = 90

    Text1.Top = 135

    Text1.Width = 3795

    Text2.Text = vbnulltring

    Text2.Height = 285

    Text2.Left = 90

    Text2.Top = 1035

    Text2.Width = 3750

    Text3.Text = vbnulltring

    Text3.Height = 285

    Text3.Left = 4095

    Text3.Top = 135

    Text3.Width = 3765

    Me.Height = 5445

    Me.Width = 11790

End Sub
NOTA: no presionen el boton de CHEQUEO hasta que yo les diga
El programa funciona de la siguiente manera en el Text1, colocan el archivo a buscar ejemplo : “prueba.txt”, en el Text2, ponemos el directorio ACTIVO en donde vamos a buscar dicho archivo, y el Text3 esta programado para buscar mas de un archivo, es decir, pongan por ejemplo “*.txt” y en list2 se les mostrara todos los .txt que se encuentren en el directorio activo.

¿Directorio Activo o Actual?, yo veo a este directorio como la carpeta a la cual se esta apuntando o se quiere realizar alguna operación que en este caso es de búsqueda. Cuando yo arranco el programa desde VB, el directorio actual es el de VB, ahora imaginemos que queremos cambiarlo, pongan por ejemplo “C:\” y cuando presionen el botón para cambiar el directorio actual, verán como cambia el directorio actual. Si el directorio no existe, no se cambia y se mantiene el directorio que estaba.

GetCurrentDirectory

Api: Declare Function GetCurrentDirectory Lib "kernel32" Alias "GetCurrentDirectoryA" (ByVal nBufferLength As Long, ByVal lpBuffer As String) As Long

Función encargada de tomar el directorio actual, como primer parámetro se le pasa la cantidad de bytes que queremos recuperar y como segundo parámetro le pasamos la cadena de tipo String donde almacenaremos dicha información.

    Dim res As Long

    cDirectory = Space(MAX_PATH)

    res = GetCurrentDirectory(MAX_PATH, cDirectory)

    Label1 = Left(cDirectory, res)

Si la función se ejecuta exitosamente, dicha función retornara la longitud en bytes.

SetCurrentDirectory

Api: Declare Function SetCurrentDirectory Lib "kernel32" Alias "SetCurrentDirectoryA" (ByVal lpPathName As String) As Long

Función utilizada para asignar un nuevo directorio actual. Toma un solo parámetro y es la dirección o “path” del nuevo directorio.

Ya explicadas estas dos funciones vamos a las funciones que se encargan de realizar la búsqueda:

FindFirstFile

Declare Function FindFirstFile Lib "kernel32" Alias "FindFirstFileA" (ByVal lpFileName As String, lpFindFileData As WIN32_FIND_DATA) As Long

Función encargada de buscar un archivo o carpeta, en la carpeta o directorio que se encuentre como activo. Esta función toma dos parámetros, como primer parámetro se toma el directorio o archivo a buscar.

Y como segundo parámetro se toma una estructura que se explica a continuación:

Private Type WIN32_FIND_DATA

    dwFileAttributes As Long ‘ Especifica los atributos del archivo

    ftCreationTime As FILETIME ‘ Especifica la hora y fecha de creación del archivo

    ftLastAccessTime As FILETIME ‘ Especifica la hora y fecha de la ultima vez que se accedió al archivo.

    ftLastWriteTime As FILETIME ‘ Especifica la hora y fecha de la ultima vez cuando se escribió el archivo.

    nFileSizeHigh As Long ‘ Especifica el “HIGH-ORDER” del tamaño del archivo

    nFileSizeLow As Long ‘ Especifica el “LOW-ORDER” del tamaño del archivo

    dwReserved0 As Long ‘ Valor reservado para el sistema

    dwReserved1 As Long ’ Valor reservado para uso futuro

    cFileName As String * MAX_PATH ‘ Nombre del archivo

    cAlternate As String * 14 ‘ Nombre alternativo del archivo

End Type

Hablemos un poco de algunos parámetros, vean en primer lugar que el tamaño del archivo se divide en dos, la parte “high” y la parte “low”, para obtener el tamaño real del archivo hay que realizar la siguiente formula:

(nFileSizeHigh * maxdword) + nFileSizeLow
El parámetro dwFileAttributes puede retornar cualquiera de las siguientes constantes:

Const FILE_ATTRIBUTE_ARCHIVE = &H20

Const FILE_ATTRIBUTE_COMPRESSED = &H800

Const FILE_ATTRIBUTE_DIRECTORY = &H10

Const FILE_ATTRIBUTE_ENCRYPTED = &H4000

Const FILE_ATTRIBUTE_HIDDEN = &H2

Const FILE_ATTRIBUTE_NORMAL = &H80

Const FILE_ATTRIBUTE_OFFLINE = &H1000

Const FILE_ATTRIBUTE_READONLY = &H1

Const FILE_ATTRIBUTE_REPARSE_POINT = &H400

Const FILE_ATTRIBUTE_SPARSE_FILE = &H200

Const FILE_ATTRIBUTE_SYSTEM = &H4

Const FILE_ATTRIBUTE_TEMPORARY = &H100

Constantes explicadas en la parte de arriba.

Ahora vamos a explicar los siguientes parámetros : ftCreationTime, ftLastAccessTime y ftLastWriteTime. Estas constantes son de otro tipo de estructura llamada FILETIME. Que es de la forma siguiente:

Private Type FILETIME

    dwLowDateTime As Long

    dwHighDateTime As Long

End Type

Es fácil, darse cuenta, que en esta estructura no hay hora, fechas, meses, días, ni nada por el estilo, aquí el tiempo esta representado por un formato, que particularmente no nos sirve. Dicho formato de tiempo es el UTC (Coordinated Universal Time), por consiguiente tenemos que transformar esta estructura en un formato mas entendible para nosotros. Y para eso esta las siguientes funciones:

FileTimeToLocalFileTime

Api: Declare Function FileTimeToLocalFileTime Lib "kernel32" (lpFileTime As FILETIME, lpLocalFileTime As FILETIME) As Long

Esta función convierte el tiempo que le pasamos como primer parámetro que posee el formato UTC, y lo convierte en el formato “local file time”, que es almacenado en el segundo parámetro “lpLocalFileTime”, que es también del tipo FILETIME. Todavía, no sirve este formato, pero es necesario aplicar esta función para después aplicar la función siguiente:

FileTimeToSystemTime

Api: Declare Function FileTimeToSystemTime Lib "kernel32" (lpFileTime As FILETIME, lpSystemTime As SYSTEMTIME) As Long

Esta función sirve para transformar del formato “local file time” en formato “system time”, y este formato si ya es un formato entendible para nosotros. Como primer parámetro  le pasamos la variable que colocamos en “lpLocalFileTime” de la función anterior. Y el resultado de dicha transformación se almacena, en el segundo parámetro de la función que es de tipo SYSTEMTIME. Dicha estructura se representa de la siguiente manera:

Private Type SYSTEMTIME

    wYear As Integer ‘ Año

    wMonth As Integer ‘ Mes

    wDayOfWeek As Integer ‘ Día de la semana, expresado en  número

    wDay As Integer ‘ Dia.

    wHour As Integer ‘ Hora

    wMinute As Integer ‘ Minuto

    wSecond As Integer ‘ Segundos
    wMilliseconds As Integer ‘ Milisegundos

End Type

Este formato si ya es entendible por nosotros y lo podemos utilizar para saber los distintos tiempos que manejan los 3 parámetros antes mencionados de la estructura WIN32_FIND_DATA.
De todas maneras cuando vayamos al código entenderán mejor.

Regresando a la función FindFirstFile,  hay que tener en cuenta que si la función encuentra algún archivo dicha función retorna un “handle” de búsqueda, en caso de fallar la búsqueda retornaría INVALID_HANDLE_VALUE.

Para verlo todo mas claro vayamos al ejemplo:

    hBusqueda = FindFirstFile(Text1.Text, wData) ‘ empiezo la búsqueda del archivo especificado en el Text1.

    If hBusqueda = INVALID_HANDLE_VALUE Then ‘ Si la búsqueda falla seria por que tal vez no se encontró.

        MsgBox "Archivo no encontrado"

        Exit Sub

    End If

    List1.AddItem wData.cFileName ‘ Muestro el nombre del archivo en el list1

    If wData.dwFileAttributes And FILE_ATTRIBUTE_ARCHIVE Then ‘ Verifico si el archivo posee cada uno de los atributos.

        List1.AddItem "Este es un archivo"

    Else

        List1.AddItem "Este no es un archivo"

    End If


.



.


.

    FileTimeToLocalFileTime wData.ftCreationTime, ftime ‘ Convierto el tiempo que me regresa la función en el tiempo “local file time”

    FileTimeToSystemTime ftime, stime ‘ Convierto “local file time” en “system time” que es el formato el cual entendemos.

    List1.AddItem " Dia = " & stime.wDay

    List1.AddItem " Dia de la semana = " & stime.wDayOfWeek

    List1.AddItem " Hora = " & stime.wHour

    List1.AddItem " Minuto = " & stime.wMinute

    List1.AddItem " Segundos = " & stime.wSecond

    List1.AddItem " Milisegundos = " & stime.wMilliseconds

    List1.AddItem " MEs = " & stime.wMonth

    List1.AddItem " Año = " & stime.wYear

    Muestro las especificaciones del tiempo en el list1.

.

.

.

    List1.AddItem "Tamaño = " & ((wData.nFileSizeHigh * maxdword) + wData.nFileSizeLow)      ‘ Calculo el tamaño del archivo

    FindClose búsqueda ‘ cierro el “handle” de búsqueda.

FindClose

Api: Declare Function FindClose Lib "kernel32" (ByVal hFindFile As Long) As Long

Función encargada de cerrar un “handle” de búsqueda que se encuentre abierto. Como único parámetro le pasamos el “handle” del archivo encontrado.

Ya sabemos como buscar un archivo, pero que sucede si queremos buscar una serie de archivos por ejemplo, queremos buscar todos los “*.bmp” que posea una carpeta para ello es importante la siguiente función:

FindNextFile

Api: Declare Function FindNextFile Lib "kernel32" Alias "FindNextFileA" (ByVal hFindFile As Long, lpFindFileData As WIN32_FIND_DATA) As Long

Esta función es la encargada de buscar el archivo “siguiente”, ya que FindFirstFile busca el primero y luego será trabajo de esta función buscar el siguiente archivo, como primer parámetro le pasamos el primer archivo encontrado, es decir, el “handle” que nos retorna FindFirstFile, y como segundo parámetro le pasamos una variable del tipo  WIN32_FIND_DATA.

Si la función es un éxito ella retorna el nuevo “handle” del archivo encontrado. En caso de fallar retorna 0. ¿Cómo saber que el error es por no haber mas archivo?.

Para ello usamos una función que no quería tratar, pero me vi obligado a usarla en este ejemplo y es:

GetLastError

Api: Declare Function GetLastError Lib "kernel32" () As Long

Esta función retorna el último error que se produjo en un hilo o “thread”. Dicha función no toma ningún parámetro y devuelve un código y dependiendo del código representa algún error.

Dichos códigos los encuentran en MSDN, en la sección “Error Codes”. En este caso el error que me interesa capturar es:

Const ERROR_NO_MORE_FILES = 18&

Vamos aclarar algo ¿Por qué no quería tocar esta función?. Esta función no se usa nada mas en archivo sino en muchas funciones de cualquier tipo, por ejemplo en las graficas se puede usar dicha función ¿Por qué usar esta función? Algunas veces cuando una función falla, puede fallar por muchas razones, por ejemplo, en algunas funciones yo he dicho que si la función falla dicha función retorna 0. ¿Pero, por que exactamente fallo? Es aquí donde GetLastError entra, ya que esta función nos proporciona de una manera mas detalla que fue lo que paso y por que fallo la función.

Por ejemplo FindNextFile retorna FALSE si falla, pero como se que falla por que no hay mas archivos o por otra cosa. Entonces es donde la función cumple un papel primordial. No quería tocar este tema ya que es MUY AMPLIO, es decir, a medida que salen nuevas versiones de Windows, los errores van aumentando, y cada función retorna errores específicos. Y en realidad este tema de errores no juega un papel primordial y la finalidad de este manual.

Si ven los ejemplos de MSDN, verán que los programadores son muy cuidadosos al momento de capturar errores, en mis ejemplos yo he sido menos cuidadoso, ya que desarrollo los ejemplos de tal manera que no fallen. Por consiguiente si desean estudiar los errores, tendrán que estudiar los códigos de los errores, y dependiendo de la función tendrán que ver el error especifico que genera dicha función. Pero ese trabajo se los dejo a ustedes.

Analicemos el código:

    Dim encontrado As Boolean ‘ Variable que controla el WHILE que se expone abajo.

    Dim hbusqueda2 As Long

    encontrado = True

    List2.Clear

    hBusqueda = FindFirstFile(Text3.Text, wData) ‘ Buscamos el primer archivo

    If hBusqueda = INVALID_HANDLE_VALUE Then ‘ Si no existe ni siquiera uno entonces nos salimos.

        MsgBox "Archivo no encontrado"

        Exit Sub

    End If

    Do While encontrado ‘ Mientras encontremos

        List2.AddItem wData.cFileName ‘ Mostramos el nombre del archivo encontrado

        If Not FindNextFile(hBusqueda, wData) Then ‘ Buscamos el siguiente, si retorna 0, que es equivalente a FALSE

            If GetLastError = ERROR_NO_MORE_FILES Then ‘ Verificamos que el error haya sido por que no había mas archivos.

                encontrado = False

            End If

        End If

    Loop ‘ Salimos del bucle.

    FindClose búsqueda ‘ Cerramos el “handle” de búsqueda.

Para los curiosos que presionaron el botón Command4, si saber lo que hacia, seguramente se habrán dado cuenta que su programa se queda como bloqueado, y seguramente tuvieron que cerrar todo VB para resolver dicha tranca. Pues déjenme decirles que el programa no estaba trancado sino mas bien estaba “esperando” que algo sucediera, esta parte va ser explicado brevemente por la razón siguiente.

Windows posee funciones sincronizadas o como MSDN las pone “Synchronization” y de este hay todo un tema, “Synchronization” es la manera de coordinar múltiples hilos o thread que se están ejecutando. Es decir, podemos tener 2 programas y través de estas funciones sincronizar ambas aplicaciones, es decir, que una aplicación haga “algo”, cuando la otra termine de hacer “algo”. Dicho tema no lo trato en este manual ya que es complejo, y no poseo muchos conocimientos respecto al tema. Pero aquellos interesados pueden buscar “Synchronization” en MSDN y leer todo un tema acerca de eso.

Pero aquí en este ejemplo use una función de las llamadas “wait function” que son funciones que dejan el programa en espera de “algo”, vamos a verlo mas claro, corran el programa y denle clic al Command4, el programa pareciera como si estuviera trancado pero no es así. Vayan ahora a “Mi Pc” y luego se meten en “C:”, estando en “C:” creen un archivo cualquiera, después de crearlo vayan al programa y verán que ya no esta trancado y ven un mensaje "EPA POR QUE ANDAS CREANDO ARCHIVOS". ASOMBROSO!!!!. Ustedes dirán ¿Cómo sabes que cree un archivo? Para responder a esa pregunta veamos la siguiente función:

FindFirstChangeNotification

Api: Declare Function FindFirstChangeNotification Lib "kernel32" Alias "FindFirstChangeNotificationA" (ByVal lpPathName As String, ByVal bWatchSubtree As Long, ByVal dwNotifyFilter As Long) As Long

Esta función sirve para ver o notificar algún cambio en un directorio. Dicha función se debe de utilizar con una función de sincronización que en este caso usamos una función de tipo “wait”. Ya hablaremos de eso adelante.

Como primer parámetro le pasamos la ruta del directorio que queremos observar. Como segundo parámetro “bWatchSubtree” le pasamos TRUE o FALSE, si es FALSE solo se monitorea el directorio que se especifico en el segundo parámetro, en caso de que sea TRUE se monitorea desde el “root” del directorio especificado, y por ultimo tenemos el  tercer parámetro que le podemos pasar las siguientes constantes, que especifican que tipo de notificación queremos capturar:

Const FILE_NOTIFY_CHANGE_FILE_NAME = &H1 ‘ Con este parámetro monitoreamos la creación, eliminación o cambio de nombre de algún archivo.

Const FILE_NOTIFY_CHANGE_ATTRIBUTES = &H4 ‘ Con este monitoreamos alguna modificación de los atributos del directorio.

Const FILE_NOTIFY_CHANGE_DIR_NAME = &H2 ‘ Monitoreamos la creación y eliminación de algún directorio.

Const FILE_NOTIFY_CHANGE_SIZE = &H8 ‘ Monitoreamos algún cambio en el tamaño del directorio.

Const FILE_NOTIFY_CHANGE_LAST_WRITE = &H10 ‘ Monitoreamos las ultimas entradas a los archivos y directorios.

Const FILE_NOTIFY_CHANGE_SECURITY = &H100 ‘ Monitoreamos algún cambio de seguridad.

Ahora esta función no sirve, si no usamos alguna función de “wait function”. Que en nuestro caso es:

WaitForSingleObject

Private Declare Function WaitForSingleObject Lib "kernel32" (ByVal hHandle As Long, ByVal dwMilliseconds As Long) As Long

NOTA: hay otras funciones que se pueden utilizar, que lo encuentran en el apartado que mencione arriba, como no vamos a explicar con detenimiento el funcionamiento de todas las funciones, nada mas me remito a explicar la que use. Que en este caso es WaitForSingleObject.
Esta función nos sirve para esperar las alteraciones que se producen en un objeto, que en nuestro caso es un directorio. Como primer parámetro se le pasa dicho objeto. Y como segundo parámetro se le pasa el tiempo en milisegundos, que queremos que este esperando. Ya que seria muy riesgoso dejarlo de manera infinita. En caso de pasarle 0, la función verifica el estado del objeto y retorna inmediatamente.

Los valores que puede retornar son:

Const WAIT_ABANDONED = ((STATUS_ABANDONED_WAIT_0) + 0)

Const WAIT_OBJECT_0 = ((STATUS_WAIT_0) + 0) ‘ Significa que la condición que pusimos a monitorear se estableció, es decir, “sucedió”.

Const WAIT_TIMEOUT = 258& ‘ Significa que la función termino por limite de tiempo.

En español, como seria entender todo esto, vayamos al código:

    hChequeo = FindFirstChangeNotification("C:\", False, FILE_NOTIFY_CHANGE_FILE_NAME) ‘ Creo un objeto en donde se monitorea a “C:\”, y lo que voy a monitorear es algún cambio en los archivos.

    Dim hObject As Long

    Dim bucle As Boolean

    bucle = True

    hObject = WaitForSingleObject(hChequeo, 50000) ‘ Creo el objeto de espera, el cual esperara que se produzcan dichos cambios en un periodo de 5 minutos. El sistema se mantendrá BLOQUEADO hasta que algún cambio ocurra o pasen los 5 minutos.

    Select Case hObject

    Case WAIT_OBJECT_0 ‘ Si hObject es WAIT_OBJECT_0 significa que hubo algún cambio, en los nombre, creación o eliminación de algún archivo en C:.

        Label2.Caption = "EPA POR QUE ANDAS CREANDO ARCHIVOS"

    Case WAIT_TIMEOUT ‘ Si entra aquí significa que el tiempo de espera termino.

        MsgBox "TIEMPO EXCEDIDO"

    End Select

    FindCloseChangeNotification hChequeo ‘ Cierro el objeto que se esta monitoreando.

FindCloseChangeNotification

Api: Declare Function FindCloseChangeNotification Lib "kernel32" (ByVal hChangeHandle As Long) As Long

Función que se encarga de cerrar el objeto creado con FindFirstChangeNotification.
Pueden ver lo interesante y asombroso de todo esto, y espero que al igual que a mi les haya gustado.

TIPOS DE UNIDADES

En este proyecto veremos como saber que tipo de unidades representan cada “letra” y el espacio disponible en cada unidad. Para el proyecto se necesitan solo 2 ListBox. Luego pegan este código:

Private Const DRIVE_CDROM = 5

Private Const DRIVE_FIXED = 3

Private Const DRIVE_NO_ROOT_DIR = 1

Private Const DRIVE_RAMDISK = 6

Private Const DRIVE_REMOTE = 4

Private Const DRIVE_REMOVABLE = 2

Private Const DRIVE_UNKNOWN = 0

Private Declare Function GetDiskFreeSpaceEx Lib "kernel32.dll" Alias "GetDiskFreeSpaceExA" (ByVal lpDirectoryName As String, ByRef lpFreeBytesAvailableToCaller As Currency, ByRef lpTotalNumberOfBytes As Currency, ByRef lpTotalNumberOfFreeBytes As Currency) As Long

Private Declare Function GetDriveType Lib "kernel32" Alias "GetDriveTypeA" (ByVal nDrive As String) As Long

Private Declare Function GetLogicalDriveStrings Lib "kernel32" Alias "GetLogicalDriveStringsA" (ByVal nBufferLength As Long, ByVal lpBuffer As String) As Long

Private Sub Form_Click()

    Dim s As String

    s = Space(255)

    Dim res As Long

    res = GetLogicalDriveStrings(255, s)

    s = Left(s, res)

    Debug.Print s

End Sub

Private Sub Form_Load()

    List1.Height = 2985

    List1.Left = 45

    List1.Top = 45

    List1.Width = 2445

    List2.Height = 2985

    List2.Left = 2655

    List2.Top = 45

    List2.Width = 4695

    Me.Height = 3870

    Me.Width = 7545

    Dim i As Integer

    For i = 97 To 122

        List1.AddItem UCase(Chr(i)) & ": " & ObtenerTipo(Chr(i))

    Next

End Sub

Function ObtenerTipo(letra As String) As String

    Dim res As Long

    res = GetDriveType(letra & ":")

    Select Case res

    Case DRIVE_CDROM

        ObtenerTipo = "CD-ROM"

    Case DRIVE_FIXED

        ObtenerTipo = "DISCO FIJO"

    Case DRIVE_UNKNOWN

        ObtenerTipo = "DESCONOCIDO"

    'Case DRIVE_NO_ROOT_DIR

    '    ObtenerTipo = "NO EXISTE DIRECTORIO ROOT"

    Case DRIVE_REMOVABLE

        ObtenerTipo = "DISCO FLEXIBLE"

    Case DRIVE_REMOTE

        ObtenerTipo = "DISCO REMOTO RED"

    Case DRIVE_RAMDISK

        ObtenerTipo = "DISCO RAM"

    End Select

End Function

Private Sub List1_Click()

    Call ObtenerTamaño(Left(List1.Text, 2))

End Sub

Sub ObtenerTamaño(letra As String)

    Dim totalBytes As Currency, totalBytesC As Currency, totalUsado As Currency, totallibres As Currency

    GetDiskFreeSpaceEx letra, totalBytesC, totalBytes, totallibres

    List2.Clear

    List2.AddItem "Total de numero de bytes = " & Format$(totalBytes * 10000, "###,###,###,##0") & " bytes"

    List2.AddItem "Total de bytes libres = " & Format$(totallibres * 10000, "###,###,###,##0") & " bytes"

    List2.AddItem "Bytes libres Habilitados = " & Format$(totalBytesC * 10000, "###,###,###,##0") & " bytes"

    List2.AddItem "Total espacio usado = " & Format$((totalBytes - totallibres) * 10000, "###,###,###,##0") & " bytes"

End Sub
Creo que si corren el programa es fácil, ver lo que hace. Por lo que vamos a ir directo a las funciones involucradas en dicho proceso.

GetDriveType

Api: Declare Function GetDriveType Lib "kernel32" Alias "GetDriveTypeA" (ByVal nDrive As String) As Long

Función utilizada para conocer que tipo de dispositivo representa alguna “ruta” o “letra”. La función retorna el tipo de dispositivo que es. Que viene determinado por las siguientes constantes:

Const DRIVE_CDROM = 5 ‘ Significa que dicha unidad es un CDROM

Const DRIVE_FIXED = 3 ‘ Significa que la unidad es un disco rígido, o disco duro

Const DRIVE_RAMDISK = 6 ‘ Significa que la unidad un disco en memoria ram. Para poner un ejemplo, es lo que hace el disco de inicio de Windows 98.

Const DRIVE_REMOTE = 4 ‘ Significa que la unidad es un disco conectado a una RED.

Const DRIVE_REMOVABLE = 2 ‘ significa que la unidad es de disco flexible, ejemplo Floppy.

Const DRIVE_UNKNOWN = 0 ‘ Significa que la unidad es desconocida.

Ahora cada vez que presionan clic en una unidad “valida”, llamo unidad valida a una “letra” que represente algún dispositivo mencionado arriba, verán en el segundo listbox, el espacio libre y ocupado del mismo. Esto se logra por la siguiente función.

GetDiskFreeSpaceEx

Api: Declare Function GetDiskFreeSpaceEx Lib "kernel32.dll" Alias "GetDiskFreeSpaceExA" (ByVal lpDirectoryName As String, ByRef lpFreeBytesAvailableToCaller As Currency, ByRef lpTotalNumberOfBytes As Currency, ByRef lpTotalNumberOfFreeBytes As Currency) As Long

Esta función sirve para saber los bytes libres y ocupados de una unidad de disco. Como primer parámetro le pasamos la unidad a verificar, como segundo parámetro pasamos una variable del tipo “currency” y dicha variable almacenara los bytes libres que están disponibles, como tercer parámetro tenemos el total de bytes que posee el dispositivo, y como último parámetro tenemos el total de bytes libres.

¿Por qué del tipo currency?. En la actualidad lo disco duros se expresan en GB, que es lo equivalente a 1.000.000.000 bytes, una variable de tipo long, no soporta dicha cifra, por consiguiente la variable en VB que puede soportar tal cifra es “currency”.

Y como podrán ver en el ejemplo estas dos funciones son fáciles de usar. Pero hay una tercera que se ejecuta cuando le dan clic en el formulario, y verán en la ventana “inmediato” todas las letras que actualmente tienen ocupadas.

GetLogicalDriveStrings

Api: Declare Function GetLogicalDriveStrings Lib "kernel32" Alias "GetLogicalDriveStringsA" (ByVal nBufferLength As Long, ByVal lpBuffer As String) As Long

Esta función se encarga de mostrar en una cadena de texto, todas las unidades lógicas que existen en la PC, como primer parámetro le pasamos la cantidad de bytes que queremos recuperar y como segundo parámetro pasamos la variable de tipo String que almacenara dicha información.

Si la función se ejecuta con éxito, esta retorna el tamaño en bytes de la cadena almacenada.

OTRAS FUNCIONES

Estas funciones que expongo a continuación, son funciones sencillas, y por esa razón no desarrolle un proyecto, sino me limito solo a explicar como funciona.

CopyFile

Api: Declare Function CopyFile Lib "kernel32" Alias "CopyFileA" (ByVal lpExistingFileName As String, ByVal lpNewFileName As String, ByVal bFailIfExists As Long) As Long

Función que copia un archivo de un lugar de origen a un lugar destino. Como primer parámetro se pasa el archivo a copiar, como segundo parámetro se tiene el archivo destino, que no tiene que llamarse igual al archivo de origen. Como tercer parámetro se le pasa TRUE o FALSE, de ser TRUE implica que si el archivo existe pues la función no hace nada, si se pone FALSE significa que en caso de existir el archivo la función lo sobrescribe.

Ejm:


Copyfile “C:\prueba.txt”, “D:\p.txt”, FALSE

MoveFile

Api: Declare Function MoveFile Lib "kernel32" Alias "MoveFileA" (ByVal lpExistingFileName As String, ByVal lpNewFileName As String) As Long

Función que sirve para renombrar a un archivo o directorio. Como primer parámetro pasamos el archivo de origen, y como segundo parámetro el archivo nuevo.

Ejm:


MoveFile “C:\prueba.txt”, “C:\p.txt”

DeleteFile

Api: Declare Function DeleteFile Lib "kernel32" Alias "DeleteFileA" (ByVal lpFileName As String) As Long

función que se encarga de eliminar a un archivo que se le pasa como primer parámetro.

GetBinaryType

Api: Declare Function GetBinaryType Lib "kernel32" Alias "GetBinaryTypeA" (ByVal lpApplicationName As String, lpBinaryType As Long) As Long

función que se encarga de determinar si un archivo es ejecutable y en caso de serlo saber que tipo de aplicación es. Dicha información se retorna en la variable que se le pasa como segundo parámetro. Es decir, después de llamar a la función la variable pasada en el segundo parámetro puede ser igual a alguna de estas constantes:

Const SCS_32BIT_BINARY = 0 ‘ El ejecutable es una aplicación basada en Windows 32Bits

Const SCS_DOS_BINARY = 1 ‘ En una aplicación de DOS

Const SCS_OS216_BINARY = 5 ‘ Es una aplicación de 16 bits basado en OS/2

Const SCS_PIF_BINARY = 3 ‘ Es una aplicación PIF que se ejecuta en DOS

Const SCS_POSIX_BINARY = 4 ‘ Es una aplicación basado en POSIX

Const SCS_WOW_BINARY = 2 ‘ Es una aplicación basado en Windows 16Bits

Si la función retorna un valor distinto de 0 implica que el archivo es un ejecutable.

GetFileAttributes

Api: Declare Function GetFileAttributes Lib "kernel32" Alias "GetFileAttributesA" (ByVal lpFileName As String) As Long

Función encargada de obtener los atributos de un archivo. Como primer parámetro se le pasa el archivo a verificar, y la función retorna los atributos del archivo. Para ver que atributos posee, aplican lo mismo que hicimos cuando usamos la estructura  WIN32_FIND_DATA      (dwFileAttributes) 

SetFileAttributes

Api: Declare Function SetFileAttributes Lib "kernel32" Alias "SetFileAttributesA" (ByVal lpFileName As String, ByVal dwFileAttributes As Long) As Long

Función utilizada para asignarle atributos a un archivo, dichos atributos con las constantes expuestas arribas que empiezan por FILE_ATTRIBUTE_..., como primer parámetro pasamos la ruta del archivo, y como segundo parámetro pasamos los nuevos atributos.

Capítulo  X. Registro de Windows y archivos .ini

En algunas ocasiones, nuestros programas necesitan alguna información para poder funcionar, dicha información podemos almacenarlas en una bases de datos o en un archivo cualquiera, pero existen estructuras cómodas y seguras para almacenar dicha información y utilizarla cuando creamos conveniente.

Todos ustedes deberán de haber escuchado o visto el “registro de Windows”, en dicho registro se encuentra configurado casi todo Windows, es decir, los colores que ven, el tipo de letra, el uso horario, las dll, los programas, los drivers,  se podría decir que toda la configuración de Windows se encuentra en el registro, un mal uso de dicho registro podría ocasionar fallas irreversibles que podrían trancar su computadora y tendrían que reinstalar Windows de nuevo. En este Capítulo  introduciremos ¿cómo es el registro de Windows?, entraremos en él teniendo siempre cuidado de lo que hagamos en su interior.

Otra forma de guardar información de manera estructurada son en los archivos .ini, dichos archivos se mantienen hoy en día, para mantener la compatibilidad con sistemas de 16 Bits, ya que, Microsoft lo que recomienda es usar el registro en vez de archivos .ini, pero estos archivos siguen siendo utilizado, ya que presentan una forma rápida y simple de acceder a datos estructurados. Quedara del programador juzgar que usar en su momento. Personalmente uso normalmente el registro ya que un archivo .ini puede ser borrado con cierta facilidad por el usuario, en cambio muy pocas veces un usuario tradicional entra en el registro de Windows, sin contar que el propio Windows respalda de manera automática el registro de Windows, es decir, Windows protege nuestra información.

En este Capítulo  veremos las funciones Api’s que permiten manipular estas dos formas de almacenar información de manera estructurada.

REGISTRO DE WINDOWS

Vamos a presionar el botón de “inicio” luego seleccionen “ejecutar” y en el formulario de ejecución coloquen “regedit”, al presionar ENTER verán que se abre como una especie de programa, el cual contiene la estructura del registro de Windows.

Antes que nada verán como una especie de árbol, que muestra las siguientes carpetas:

HKEY_CLASSES_ROOT 

HKEY_CURRENT_CONFIG 

HKEY_CURRENT_USER 

HKEY_DYN_DATA

HKEY_LOCAL_MACHINE 

HKEY_PERFORMANCE_DATA  (solo en Windows NT)
HKEY_USERS 

NOTA: mencione “carpetas”, en realidad en la documentación de Microsoft no se refieren a lo mencionado como carpeta sino como “key” o “clave”, personalmente en las explicaciones voy a usar “carpeta” ya que es la figurita que posee como icono. Y considero mas cómodo este termino que el termino “key”, pero esto lo menciono para aquellos que lean la documentación oficial. Las carpetas que contiene cada carpeta las llamaré “sub-carpetas” que seria lo equivalente a “subkey”.

HKEY_CLASSES_ROOT 

Esta carpeta contiene en pocas palabras, la configuración de las extensiones de archivos, objetos OLE, etc, es decir, ustedes se darán cuenta que un archivo de Word, posee un icono particular que lo identifica, y cada vez que le dan dobleclick abre inmediatamente el programa Word, dicha configuración se encuentra en esta sección. No voy a entrar en detalle en como se hace este proceso ya que eso no es tema de este manual, mi misión será enseñarles como “manipular” información en el registro.

HKEY_CURRENT_CONFIG 

En esta sección podemos ver la configuración actual de algunos dispositivos.

HKEY_CURRENT_USER 

En esta sección vemos la configuración del usuario actual, es decir, aquí podemos encontrar información sobre su wallpaper, colores personalizados, etc.

HKEY_USERS 

En esta sección vemos al igual que la explicación anterior la configuración pero ahora no de un solo usuario sino de los usuario configurados por el sistema.

¿Cuál es la diferencia entre HKEY_CURRENT_USER y HKEY_USERS? Todos sabemos que en Windows podemos configurar distintos usuarios, cada usuario puede entrar con su clave e inicie una especie de sesión, un usuario A por ejemplo puede tener un wallpaper de una playa, y el usuario B puede tener un wallpaper de una montaña, cuando cada usuario inicia su sesión, Windows busca su configuración en HKEY_USERS y la pega en HKEY_CURRENT_USER, con esto quiero que vean que la información que posee HKEY_CURRENT_USER es variable y puede cambiar, esto lo tienen que tener en cuenta al momento que quieran guardar alguna información. Inclusive lo que hay en HKEY_CURRENT_USER se encuentra repetido en HKEY_USER, cuando el usuario cierra su sesión, Windows guarda la información en HKEY_USERS.

HKEY_LOCAL_MACHINE 

Aquí podemos encontrar información acerca de nuestra computadora, es decir, me refiero a la parte física y también a la parte de software.

Aquí mencione un resumen de cómo son mas o menos las carpetas principales del registro de Windows, dejo en claro que las carpetas que se creen tienen que ser dentro de alguna que explique arriba, es decir, no pueden crear una carpeta en el ROOT del registro. Y tampoco pueden borrar ninguna de las que menciones.

Para este proyecto se necesitan 10 botones, 2 listbox y una etiqueta o label. Luego peguen este código en el formulario:

Private Type SECURITY_ATTRIBUTES

    nLength As Long

    lpSecurityDescriptor As Long

    bInheritHandle As Long

End Type

Private Type FILETIME

    dwLowDateTime As Long

    dwHighDateTime As Long

End Type

Const ERROR_NO_MORE_ITEMS = 259&

Private Const REG_BINARY = 3

Private Const REG_DWORD = 4

Private Const REG_DWORD_BIG_ENDIAN = 5

Private Const REG_DWORD_LITTLE_ENDIAN = 4

Private Const REG_EXPAND_SZ = 2

Private Const REG_LINK = 6

Private Const REG_MULTI_SZ = 7

Private Const REG_NONE = 0

Private Const REG_RESOURCE_LIST = 8

Private Const REG_SZ = 1

Private Const ERROR_SUCCESS = 0&

Private Const HKEY_CLASSES_ROOT = &H80000000

Private Const HKEY_CURRENT_CONFIG = &H80000005

Private Const HKEY_CURRENT_USER = &H80000001

Private Const HKEY_DYN_DATA = &H80000006

Private Const HKEY_LOCAL_MACHINE = &H80000002

Private Const HKEY_PERF_ROOT = HKEY_LOCAL_MACHINE

Private Const HKEY_PERFORMANCE_DATA = &H80000004

Private Const HKEY_USERS = &H80000003

Private Const REG_OPTION_BACKUP_RESTORE = 4

Private Const REG_OPTION_CREATE_LINK = 2

Private Const REG_OPTION_NON_VOLATILE = 0

Private Const REG_OPTION_OPEN_LINK = (&H8)

Private Const REG_OPTION_RESERVED = 0

Private Const REG_OPTION_VOLATILE = 1

Private Const READ_CONTROL = &H20000

Private Const STANDARD_RIGHTS_ALL = &H1F0000

Private Const STANDARD_RIGHTS_READ = (READ_CONTROL)

Private Const STANDARD_RIGHTS_WRITE = (READ_CONTROL)

Private Const SYNCHRONIZE = &H100000

Private Const KEY_QUERY_VALUE = &H1

Private Const KEY_SET_VALUE = &H2

Private Const KEY_CREATE_LINK = &H20

Private Const KEY_ENUMERATE_SUB_KEYS = &H8

Private Const KEY_CREATE_SUB_KEY = &H4

Private Const KEY_NOTIFY = &H10

Private Const KEY_READ = ((STANDARD_RIGHTS_READ Or KEY_QUERY_VALUE Or KEY_ENUMERATE_SUB_KEYS Or KEY_NOTIFY) And (Not SYNCHRONIZE))

Private Const KEY_ALL_ACCESS = ((STANDARD_RIGHTS_ALL Or KEY_QUERY_VALUE Or KEY_SET_VALUE Or KEY_CREATE_SUB_KEY Or KEY_ENUMERATE_SUB_KEYS Or KEY_NOTIFY Or KEY_CREATE_LINK) And (Not SYNCHRONIZE))

Private Const KEY_EXECUTE = (KEY_READ)

Private Const KEY_WRITE = ((STANDARD_RIGHTS_WRITE Or KEY_SET_VALUE Or KEY_CREATE_SUB_KEY) And (Not SYNCHRONIZE))

Private Const REG_CREATED_NEW_KEY = &H1

Private Const REG_OPENED_EXISTING_KEY = &H2

Private Declare Function RegCreateKeyEx Lib "advapi32.dll" Alias "RegCreateKeyExA" (ByVal hKey As Long, ByVal lpSubKey As String, ByVal Reserved As Long, ByVal lpClass As String, ByVal dwOptions As Long, ByVal samDesired As Long, lpSecurityAttributes As Any, phkResult As Long, lpdwDisposition As Long) As Long

Private Declare Function RegDeleteKey Lib "advapi32.dll" Alias "RegDeleteKeyA" (ByVal hKey As Long, ByVal lpSubKey As String) As Long

Private Declare Function RegDeleteValue Lib "advapi32.dll" Alias "RegDeleteValueA" (ByVal hKey As Long, ByVal lpValueName As String) As Long

Private Declare Function RegEnumKeyEx Lib "advapi32.dll" Alias "RegEnumKeyExA" (ByVal hKey As Long, ByVal dwIndex As Long, ByVal lpName As String, lpcbName As Long, lpReserved As Long, ByVal lpClass As String, lpcbClass As Long, lpftLastWriteTime As Any) As Long

Private Declare Function RegEnumValue Lib "advapi32.dll" Alias "RegEnumValueA" (ByVal hKey As Long, ByVal dwIndex As Long, ByVal lpValueName As String, lpcbValueName As Long, ByVal lpReserved As Long, lpType As Long, lpData As Any, lpcbData As Long) As Long

Private Declare Function RegOpenKeyEx Lib "advapi32.dll" Alias "RegOpenKeyExA" (ByVal hKey As Long, ByVal lpSubKey As String, ByVal ulOptions As Long, ByVal samDesired As Long, phkResult As Long) As Long

Private Declare Function RegQueryValueEx Lib "advapi32.dll" Alias "RegQueryValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal lpReserved As Long, lpType As Long, lpData As Any, lpcbData As Long) As Long

Private Declare Function RegSetValueEx Lib "advapi32.dll" Alias "RegSetValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal Reserved As Long, ByVal dwType As Long, lpData As Any, ByVal cbData As Long) As Long

Private Declare Function RegCloseKey Lib "advapi32.dll" (ByVal hKey As Long) As Long

Dim resultado As Long

Dim hKey As Long

Private Sub Command1_Click()

    RegCreateKeyEx HKEY_LOCAL_MACHINE, "Software\MiPrueba", 0, "REG_DWORD", REG_OPTION_NON_VOLATILE, KEY_ALL_ACCESS, ByVal 0&, hKey, resultado

    If resultado And REG_CREATED_NEW_KEY Then

        Label1 = "Se ha creado la nueva clave o key"

    Else

        Label1 = "Ya existe la clave o key"

    End If

    RegCloseKey hKey

End Sub

Private Sub Command2_Click()

    RegCreateKeyEx HKEY_LOCAL_MACHINE, "Software\MiPrueba\carpeta1", 0, "PERSONAL", REG_OPTION_NON_VOLATILE, KEY_ALL_ACCESS, ByVal 0&, hKey, resultado

    RegCloseKey hKey

    RegCreateKeyEx HKEY_LOCAL_MACHINE, "Software\MiPrueba\carpeta2", 0, "DATOS", REG_OPTION_NON_VOLATILE, KEY_ALL_ACCESS, ByVal 0&, hKey, resultado

    RegCloseKey hKey

    RegCreateKeyEx HKEY_LOCAL_MACHINE, "Software\MiPrueba\carpeta3", 0, "GENERAL", REG_OPTION_NON_VOLATILE, KEY_ALL_ACCESS, ByVal 0&, hKey, resultado

    RegCloseKey hKey

End Sub

Private Sub Command3_Click()

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba\carpeta1", 0, KEY_ALL_ACCESS, hKey)

    If resultado = ERROR_SUCCESS Then

        Dim cadena As String

        Dim cadena2 As String

        cadena = "Datos tipo string"

        RegSetValueEx hKey, "", 0, REG_SZ, ByVal cadena, Len(cadena)

        RegSetValueEx hKey, "Valor1", 0, REG_SZ, ByVal cadena, Len(cadena)

        RegSetValueEx hKey, "Valor2", 0, REG_BINARY, 65535, 4

        RegSetValueEx hKey, "Valor3", 0, REG_DWORD, 2532, 4

    Else

        Label1 = "error al abrir la clave o key"

    End If

    RegCloseKey hKey

End Sub

Private Sub Command4_Click()

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba", 0, KEY_ALL_ACCESS, hKey)

    'resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software", 0, KEY_ALL_ACCESS, hKey)

    Dim nombre As String

    Dim maximo As Long

    Dim nombrec As String

    Dim maximoc As Long

    Dim cnt As Long

    maximo = 255

    nombre = Space(maximo)

    maximoc = 255

    nombrec = Space(maximo)

    If resultado = ERROR_SUCCESS Then

        While RegEnumKeyEx(hKey, cnt, nombre, maximo, ByVal 0&, nombrec, maximoc, ByVal 0&) <> ERROR_NO_MORE_ITEMS

            List1.AddItem Left$(nombre, maximo) & " clase = " & nombrec

            cnt = cnt + 1

            maximo = 255

            nombre = Space(maximo)

            maximoc = 255

            nombrec = Space(maximo)

        Wend

    End If

    RegCloseKey hKey

End Sub

Private Sub Command5_Click()

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba\Carpeta1", 0, KEY_ALL_ACCESS, hKey)

    Dim nombre As String

    Dim maximo As Long

    Dim maximo2 As Long

    Dim tipo As Long

    Dim valor As String

    Dim cnt As Long

    maximo = 255

    nombre = Space(maximo)

    maximo2 = 255

    valor = Space(maximo2)

    If resultado = ERROR_SUCCESS Then

        While RegEnumValue(hKey, cnt, nombre, maximo, ByVal 0&, tipo, ByVal valor, maximo2) <> ERROR_NO_MORE_ITEMS

            If tipo = REG_SZ Then

                List2.AddItem Left$(nombre, maximo) & " valor = " & Left$(valor, maximo2)

            End If

            cnt = cnt + 1

            maximo = 255

            nombre = Space(maximo)

            maximo2 = 255

            valor = Space(maximo2)

        Wend

    End If

    RegCloseKey hKey

End Sub

Private Sub Command6_Click()

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba\Carpeta1", 0, KEY_ALL_ACCESS, hKey)

    resultado = RegDeleteValue(hKey, "Valor1")

    If resultado = ERROR_SUCCESS Then

        Label1 = "Se borro Valor1"

    Else

        Label1 = "Error"

    End If

    RegCloseKey hKey

End Sub

Private Sub Command7_Click()

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba", 0, KEY_ALL_ACCESS, hKey)

    resultado = RegDeleteKey(hKey, "Carpeta2")

    If resultado = ERROR_SUCCESS Then

        Label1 = "Se borro carpeta"

    Else

        Label1 = "Error"

    End If

    RegCloseKey hKey

End Sub

Private Sub Command8_Click()

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba\carpeta1", 0, KEY_ALL_ACCESS, hKey)

    Dim maximo2 As Long

    Dim valor As String

    maximo2 = 255

    valor = Space(maximo2)

    RegQueryValueEx hKey, "Valor1", 0&, REG_SZ, ByVal valor, maximo2

    Label1 = Left$(valor, maximo2)

    RegCloseKey hKey

End Sub

Private Sub Command9_Click()

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba\carpeta1", 0, KEY_ALL_ACCESS, hKey)

    Dim valor As Long

    RegQueryValueEx hKey, "Valor3", 0&, REG_DWORD, valor, 4

    Label1 = valor

    RegCloseKey hKey

End Sub

Private Sub Command10_Click()

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba\carpeta1", 0, KEY_ALL_ACCESS, hKey)

    Dim valor As Long

    RegQueryValueEx hKey, "Valor2", 0&, REG_BINARY, valor, 4

    Label1 = valor

    RegCloseKey hKey

End Sub

Private Sub Form_Load()

    Command1.Caption = "Crear 'Key'"

    Command1.Height = 465

    Command1.Left = 45

    Command1.Top = 45

    Command1.Width = 2355

    Command2.Caption = "Crear 'SubKey'"

    Command2.Height = 465

    Command2.Left = 45

    Command2.Top = 585

    Command2.Width = 2355

    Command3.Caption = "Crear 'Value'"

    Command3.Height = 465

    Command3.Left = 45

    Command3.Top = 1125

    Command3.Width = 2355

    Command4.Caption = "Enumerar 'Keys'"

    Command4.Height = 465

    Command4.Left = 45

    Command4.Top = 1710

    Command4.Width = 2355

    Command5.Caption = "Enumerar 'Values'"

    Command5.Height = 465

    Command5.Left = 45

    Command5.Top = 2295

    Command5.Width = 2355

    Command6.Caption = "Eliminar 'Values'"

    Command6.Height = 465

    Command6.Left = 45

    Command6.Top = 2880

    Command6.Width = 2355

    Command7.Caption = "Eliminar 'Keys'"

    Command7.Height = 465

    Command7.Left = 45

    Command7.Top = 3420

    Command7.Width = 2355

    Command8.Caption = "Recuperar Cadena"

    Command8.Height = 465

    Command8.Left = 2565

    Command8.Top = 3960

    Command8.Width = 1635

    Command9.Caption = "Recuperar Dword"

    Command9.Height = 465

    Command9.Left = 4275

    Command9.Top = 3960

    Command9.Width = 1635

    Command10.Caption = "Recuperar Binario"

    Command10.Height = 465

    Command10.Left = 5985

    Command10.Top = 3960

    Command10.Width = 1635

    List1.Height = 3765

    List1.Left = 2565

    List1.Top = 45

    List1.Width = 2640

    List2.Height = 3765

    List2.Left = 5280

    List2.Top = 45

    List2.Width = 2385

    Label1.Caption = vbNullString

    Label1.Height = 465

    Label1.Left = 90

    Label1.Top = 3960

    Label1.Width = 2310

    Me.Height = 5040

    Me.Width = 7875

End Sub
Corren la aplicación, pero mantengan el programa “regedit” abierto. Primero presionen el botón “Crear key”, luego de haberlo presionado se van al programa regedit, y vayan a este ubicación “HKEY_LOCAL_MACHINE\Software”, en el listado busquen la carpeta “MiPrueba” se darán cuenta que no tiene subcarpetas, en caso de no verlo presionen F5 para refrescar el registro.

Ahora presionen el botón “Crear Subkey”, vayan al programa “regedit” y con la carpeta “MiPrueba” seleccionada presionen F5 y se darán cuenta que ahora hay 3 subcarpetas. “carpeta1”, “carpeta2” y “carpeta3”, dichas carpetas no tiene información todavía. Ahora presionen “Crear Value” y luego presionan F5, y se darán cuenta que en “carpeta1”, ahora hay 3 valores: “Valor1” que es de tipo String, “Valor2” que es de tipo binario y “Valor3” que es de tipo DWORD, Dword es una variable de 32 Bits, o 4 byte, lo que seria equivalente a una variable del tipo long.

Al presionar “Enumerar Keys” se muestra en el listbox1 las carpetas que contiene “MiPrueba”, y al presionar “Enumerar Values” se muestra en el listbox2 los datos almacenados en la “carpeta1”. Los demás botones serán explicados cuando explique las funciones involucradas.

RegCreateKeyEx

Declare Function RegCreateKeyEx Lib "advapi32.dll" Alias "RegCreateKeyExA" (ByVal hKey As Long, ByVal lpSubKey As String, ByVal Reserved As Long, ByVal lpClass As String, ByVal dwOptions As Long, ByVal samDesired As Long, lpSecurityAttributes As Any, phkResult As Long, lpdwDisposition As Long) As Long

Con esta función creamos o abrimos una “carpeta” o key, dejo en claro algo, “subcarpeta” y “carpeta” son lo mismo, al momento de crearlas, es decir, llamamos subcarpetas a las carpetas que están subordinadas por otras, es decir, la función para crear cualquier carpeta es esta. Esto lo menciono por si esperan una función para crear los “subkeys” o “subcarpetas”, función que no existe ya que es esta misma.

“hkey” =  le podemos pasar una de las siguientes constantes:

Private Const HKEY_CLASSES_ROOT = &H80000000

Private Const HKEY_CURRENT_CONFIG = &H80000005

Private Const HKEY_CURRENT_USER = &H80000001

Private Const HKEY_DYN_DATA = &H80000006

Private Const HKEY_LOCAL_MACHINE = &H80000002

Private Const HKEY_PERF_ROOT = HKEY_LOCAL_MACHINE

Private Const HKEY_PERFORMANCE_DATA = &H80000004

Private Const HKEY_USERS = &H80000003

Es decir, con estas constantes configuramos en que carpeta del ROOT estará contenida nuestra nueva carpeta.

“lpSubKey” = le pasamos la ubicación de nuestra nueva carpeta. El tercer parámetro “Reserved” siempre se la pasa 0.

“lpClass” = es una forma de identificar nuestras carpetas, en realidad es de tipo String e identifica el tipo de carpeta, !!en español!!, podemos tener 10 carpetas e identificar las primeras 5 con datos del programa, y las otras 5 datos del usuario, entonces podemos poner en este parámetro en las primeras 5 el valor de “PROGRAM” y las otras 5 el valor de “USER”, en cierta forma identificamos que tipo de carpeta es. Lo que puede servir al momento de consultarlas.

“dwOptions” = puede tomar las siguientes constantes:

Private Const REG_OPTION_BACKUP_RESTORE = 4 
Private Const REG_OPTION_NON_VOLATILE = 0 ‘ La carpeta o key es no volátil, esta opción es por defecto. La información es almacenada en un archivo y es conservada cuando el sistema es reiniciado.

Private Const REG_OPTION_VOLATILE = 1 ‘ En Windows 95 es ignorado. En NT, implica que la información es almacenada en memoria y no es conservada cuando el sistema es reiniciado.

“samDesired” = especificamos los permisos de la nueva carpeta, en esto parámetro podemos usar el OR para especificar mas de una constante:

KEY_READ = ((STANDARD_RIGHTS_READ Or KEY_QUERY_VALUE Or KEY_ENUMERATE_SUB_KEYS Or KEY_NOTIFY) And (Not SYNCHRONIZE)) ‘ Con esto especificamos que nuestra carpeta esta siendo abierta o creada solo con permiso para leer, en pocas palabras no podemos grabar en dicha carpeta.

KEY_ALL_ACCESS = ((STANDARD_RIGHTS_ALL Or KEY_QUERY_VALUE Or KEY_SET_VALUE Or KEY_CREATE_SUB_KEY Or KEY_ENUMERATE_SUB_KEYS Or KEY_NOTIFY Or KEY_CREATE_LINK) And (Not SYNCHRONIZE)) ‘ Con este parametro tenemos acceso total en nuestra carpeta.

KEY_EXECUTE = (KEY_READ)  ‘Es lo equivalente a tener permiso solo para leer.

KEY_WRITE = ((STANDARD_RIGHTS_WRITE Or KEY_SET_VALUE Or KEY_CREATE_SUB_KEY) And (Not SYNCHRONIZE)) ‘ Con esta constante tenemos permiso para escribir en nuestra carpeta.

KEY_CREATE_SUB_KEY = &H4 ‘ Tenemos permiso para crear otras capetas o subcarpetas. Que estén contenidas en la carpeta creada o abierta.

KEY_ENUMERATE_SUB_KEYS = &H8 ‘ Tenemos permiso para enumerar las carpetas o subcarpetas de una carpeta abierta.

KEY_QUERY_VALUE = &H1 ‘ Tenemos permiso para consultar los valores que contenga la carpeta.

KEY_SET_VALUE = &H2 ‘ Tenemos permiso para crear o alterar los valores de una carpeta.

“lpSecurityAttributes” = en la función esta declarada del tipo “Any” dicha declaración esta hecha a propósito, en realidad este parámetro es del tipo SECURITY_ATTRIBUTES:

Private Type SECURITY_ATTRIBUTES

    nLength As Long

    lpSecurityDescriptor As Long

    bInheritHandle As Long

End Type
Lo que sucede es que como no vamos a usar dicha estructura, cambie la declaración en la función para que no tuviera problema al momento de ejecutarla. Dicha estructura posee mas funcionalidad en NT. 

“phkResult” = es quien va a tener el “handle” de la carpeta o “key” creada o abierta. Dicha variable es la que utilizamos en la manipulación de dicha carpeta.

“lpdwDisposition” = parámetro el cual después de ejecutar la función puede tener 2 valores:

Private Const REG_CREATED_NEW_KEY = &H1 ‘ Esta constante determina que la carpeta fue creada.

Private Const REG_OPENED_EXISTING_KEY = &H2 ‘ Esta constante especifica que ya existía la carpeta.

RegCloseKey

Api: Declare Function RegCloseKey Lib "advapi32.dll" (ByVal hKey As Long) As Long

Esta función cierra el “handle” de una carpeta que haya sido abierta. Como único parámetro se le pasa dicho “handle”.

En nuestro proyecto estas dos funciones la usamos en el primer botón como se indica:

RegCreateKeyEx HKEY_LOCAL_MACHINE, "Software\MiPrueba", 0, "REG_DWORD", REG_OPTION_NON_VOLATILE, KEY_ALL_ACCESS, ByVal 0&, hKey, resultado

    Creamos una carpeta en HKEY_LOCAL_MACHINE, en la carpeta “Software” y la carpeta nueva se llama “MiPrueba”, la identifico como “REG_DWORD” pero puedes poner el nombre que quiera, le especifico que sea volátil, con todos los accesos, la variable “hkey” sirve para almacenar el “handle”  de la carpeta

    If resultado And REG_CREATED_NEW_KEY Then ‘ Verificamos si fue creado con existo o si ya estaba creada.

        Label1 = "Se ha creado la nueva clave o key"

    Else

        Label1 = "Ya existe la clave o key"

    End If

    RegCloseKey hKey ‘ Cerramos el “handle”.

RegOpenKeyEx

Api: Declare Function RegOpenKeyEx Lib "advapi32.dll" Alias "RegOpenKeyExA" (ByVal hKey As Long, ByVal lpSubKey As String, ByVal ulOptions As Long, ByVal samDesired As Long, phkResult As Long) As Long

Esta función se utiliza para abrir una carpeta o “key” ya existente.

“hkey” = Toma las mismas constantes que la función RegCreateKeyEx.

“lpSubKey” = Aquí se especifica la ubicación de la carpeta o “key” que se quiere abrir.

“ulOptions” = reservado, siempre se le paso 0.

“samDesired” = Tomas las mismas constantes que la función RegCreateKeyEx.

“phkResult” = La variable que se le pase en este parámetro tomara el “handle”, de la carpeta abierta.

La función en caso de abrir con éxito la carpeta retornara ERROR_SUCCESS.

RegSetValueEx

Api: Declare Function RegSetValueEx Lib "advapi32.dll" Alias "RegSetValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal Reserved As Long, ByVal dwType As Long, lpData As Any, ByVal cbData As Long) As Long

Esta función sirve para asignarles valores a la carpeta, que fue abierta con RegOpenKeyEx, el termino “valores” en la ayuda de MSDN se le denomina “data”.

“hkey” = aquí se le pasa el “handle” de la carpeta abierta, en pocas palabras la variable que usaron en el parámetro de “phkResult” en la función RegOpenKeyEx, es la variable que colocan aquí.

“lpValueName” = aquí colocamos el nombre del valor o “data”.

“reserved” = reservado, se le pasa siempre 0.

“dwType” = especificamos el tipo de dato que queremos almacenar, puede tomar las siguientes constantes:

Private Const REG_BINARY = 3 ‘ Especifica que el valor es binario

Private Const REG_DWORD = 4 ‘ Especifica que el valor es de tipo long o de 32 bits.

Private Const REG_DWORD_BIG_ENDIAN = 5 ‘ Especifica un valor del tipo long, pero que se organiza de la siguiente manera: 0x12345678 = (0x12 0x34 0x56 0x78)

Private Const REG_DWORD_LITTLE_ENDIAN = 4 ‘ Especifica un valor del tipo long, pero que se organiza de la siguiente manera: 0x12345678 = (0x78 0x56 0x34 0x12)
Private Const REG_EXPAND_SZ = 2 ‘ Usado para cadenas de texto que contienen que referencia a variables globales o “environment variables”.

Private Const REG_LINK = 6 ‘ Unicode simbolic link.

Private Const REG_MULTI_SZ = 7 ‘  Especifica un arreglo de cadena.

Private Const REG_NONE = 0 ‘  No define ningún valor

Private Const REG_RESOURCE_LIST = 8 ‘ Especifica una lista de recursos de un driver.

Private Const REG_SZ = 1 ‘ Especifica que el valor es del tipo texto

“lpData” = aquí pasamos el valor.

“cbData” = pasamos el tamaño del valor.

En el código lo usamos de esta manera:

    resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba\carpeta1", 0, KEY_ALL_ACCESS, hKey) ‘ Abrimos la carpeta “carpeta1”, con todos los accesos.

    If resultado = ERROR_SUCCESS Then ‘ Si tenemos éxito.

        Dim cadena As String

        Dim cadena2 As String

        cadena = "Datos tipo string"

        RegSetValueEx hKey, "", 0, REG_SZ, ByVal cadena, Len(cadena) ‘  Guardo la cadena en el renglón de “valor predeterminado”

        RegSetValueEx hKey, "Valor1", 0, REG_SZ, ByVal cadena, Len(cadena) ‘ Creo un nuevo valor llamado “valor1” y le paso también la cadena.

        RegSetValueEx hKey, "Valor2", 0, REG_BINARY, 65535, 4 ‘ Especifico un valor binario que conste de 4 bytes.

        RegSetValueEx hKey, "Valor3", 0, REG_DWORD, 2532, 4 ‘ Creamos un valor de tipo Dword, su tamaño siempre será de 4 bytes, BINARY puede ser mas grande o mas pequeño, no se confundan por que se vean de igual tamaño.

    Else

        Label1 = "error al abrir la clave o key"

    End If

    RegCloseKey hKey ‘  Cierro el “handle”.

RegEnumKeyEx

Api: Declare Function RegEnumKeyEx Lib "advapi32.dll" Alias "RegEnumKeyExA" (ByVal hKey As Long, ByVal dwIndex As Long, ByVal lpName As String, lpcbName As Long, lpReserved As Long, ByVal lpClass As String, lpcbClass As Long, lpftLastWriteTime As Any) As Long

Esta función sirve para enumerar las subcarpetas o subkeys de una carpeta ya abierta.

“hkey” = aquí se le pasa el “handle” de la carpeta abierta, en pocas palabras la variable que usaron en el parámetro “phkResult” en la función RegOpenKeyEx, es la variable que colocan aquí.

“dwIndex” = aquí colocamos el índice de la carpeta a recuperar.

“lpName” = variable que almacenara el nombre de la carpeta.

“lpcbName” = aquí especificamos el tamaño en bytes de la cadena “lpName” que queremos recuperar.

“lpReserved” = reservado, siempre pondrán 0.

“lpClass” = variable que almacenara el nombre de la clase.

“lpcbClass” = aquí especificamos el tamaño en bytes de la cadena “lpClass” que queremos recuperar.

“lpftLastWriteTime” = aunque en la función esta variable es del tipo ANY originalmente se le pasa una estructura que no vamos a usar en este momento que es:

Private Type FILETIME

    dwLowDateTime As Long ‘Especifica la parte baja o 32 Bits del “file time”.

    dwHighDateTime As Long ‘Especifica la parte alta o 32 bits del “file time”.

End Type

Estructura explicada en la sección de Archivos. 

En nuestro caso no hay necesidad de usar dicha información.

En el código el uso de la función se representa de la siguiente manera:

resultado = RegOpenKeyEx(HKEY_LOCAL_MACHINE, "Software\MiPrueba", 0, KEY_ALL_ACCESS, hKey) ‘Abrimos nuestra carpeta ya creada.

.


Declaramos variables


.    

maximo = 255 ‘Especificamos la cantidad de bytes que queremos recuperar.

nombre = Space(maximo) ‘Llenamos con espacios en blanco la variable que almacenara la información-.

.


.    

If resultado = ERROR_SUCCESS Then

        While RegEnumKeyEx(hKey, cnt, nombre, maximo, ByVal 0&, nombrec, maximoc, ByVal 0&) <> ERROR_NO_MORE_ITEMS ‘Mientras existan subcarpetas.

            List1.AddItem Left$(nombre, maximo) & " clase = " & nombrec ‘Añadimos el nombre de la carpeta y la clase a la cual pertenece, dejo en claro que en Windows 98 no me retorno el nombre de la clase pero en Windows 2000 si lo retorna.

            cnt = cnt + 1 ‘Muevo el índice “dwIndex”


‘Reconfiguro las variables.            

        Wend

    End If

    RegCloseKey hKey ‘Cierro la carpeta.

Hay que destacar que después que la función RegEnumKeyEx se ejecuta, el valor que posee la variable “máximo” es la cantidad de caracteres que posee la cadena. Lo mismo se aplica para “maximoc”, con la función Left, lo que estoy haciendo es desechando los espacios en blancos sobrantes. Es esa la razón del por que “reconfiguro las variables”.

RegEnumValue

Api: Declare Function RegEnumValue Lib "advapi32.dll" Alias "RegEnumValueA" (ByVal hKey As Long, ByVal dwIndex As Long, ByVal lpValueName As String, lpcbValueName As Long, ByVal lpReserved As Long, lpType As Long, lpData As Any, lpcbData As Long) As Long

Función que sirve para enumerar o retornar todos los valores que puede poseer una carpeta o key que se encuentre abierto. Es muy parecido su uso a la función RegEnumKeyEx.
“hkey” = aquí se le pasa el “handle” de la carpeta abierta, en pocas palabras la variable que usaron en el parámetro de “phkResult” en la función RegOpenKeyEx, es la variable que colocan aquí.

“dwIndex” = aquí colocamos el índice del valor a recuperar.

“lpValueName” = variable que almacenara el nombre del valor o data.

“lpcbValueName” = aquí especificamos el tamaño en bytes de la cadena “lpValueName” que queremos recuperar.

“lpReserved” = reservado, siempre pondrán 0.

“lpType” = variable que almacena el tipo de valor. Puede tomar los mismos valores que los que toma la variable “dwType” en la función RegSetValueEx.

“lpData” = variable que contiene la información de dicho valor.

“lpcbData” = aquí especificamos el tamaño en bytes de la información que queremos recuperar en “lpData”, este tamaño dependerá de la “data” a recuperar, si es DWORD tendremos que pasar el valor de 4. En caso de ser una cadena tendremos que pasar el tamaño en bytes de lo que queremos recuperar. Y así, dependiendo del tipo de dato a recuperar. En nuestro caso recuperamos los datos del tipo REG_SZ.
En el código el uso de la función se representa de la siguiente manera:


.


.


.

If resultado = ERROR_SUCCESS Then

        While RegEnumValue(hKey, cnt, nombre, maximo, ByVal 0&, tipo, ByVal valor, maximo2) <> ERROR_NO_MORE_ITEMS

            If tipo = REG_SZ Then ‘ Tomamos los valores solo del tipo RG_SZ. 

                List2.AddItem Left$(nombre, maximo) & " valor = " & Left$(valor, maximo2)

            End If

            cnt = cnt + 1 ‘‘Muevo el índice “dwIndex”


‘Reconfiguro las variables.            

        Wend

    End If

Como pueden ver su uso es muy similar a RegEnumKeyEx. 
RegDeleteValue

Api: Declare Function RegDeleteValue Lib "advapi32.dll" Alias "RegDeleteValueA" (ByVal hKey As Long, ByVal lpValueName As String) As Long

Función utilizada para eliminar un valor o data de cierta carpeta o key que este abierto.

“hkey” = aquí se le pasa el “handle” de la carpeta abierta, en pocas palabras la variable que usaron en el parámetro de “phkResult” en la función RegOpenKeyEx, es la variable que colocan aquí.

“lpValueName” = nombre del valor o data a eliminar.

RegDeleteKey

Api: Declare Function RegDeleteKey Lib "advapi32.dll" Alias "RegDeleteKeyA" (ByVal hKey As Long, ByVal lpSubKey As String) As Long

Función que elimina una subcarpeta o “subkey” del registro.

“hkey” = aquí se le pasa el “handle” de la carpeta abierta, en pocas palabras la variable que usaron en el parámetro de “phkResult” en la función RegOpenKeyEx, es la variable que colocan aquí.

“lpSubKey” = nombre de la subcarpeta contenida en “hkey” que se quiera eliminar.

RegQueryValueEx

Api: Declare Function RegQueryValueEx Lib "advapi32.dll" Alias "RegQueryValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal lpReserved As Long, lpType As Long, lpData As Any, lpcbData As Long) As Long

Función utilizada para consultar la información de cierto valor o “data”.

“hkey” = aquí se le pasa el “handle” de la carpeta abierta, en pocas palabras la variable que usaron en el parámetro de “phkResult” en la función RegOpenKeyEx, es la variable que colocan aquí.

“lpValueName” = nombre del valor o data.

“lpReserved” = reservado, siempre pondrán 0.

“lpType” = variable que almacena el tipo de valor. Puede tomar los mismos valores que los que toma la variable “dwType” en la función RegSetValueEx.

“lpData” = variable que contiene la información de dicho valor.

“lpcbData” = aquí especificamos el tamaño en bytes de la información que queremos recuperar en “lpData”, este tamaño dependerá de la “data” a recuperar si es DWORD tendremos que pasar el valor de 4. En caso de ser una cadena tendremos que pasar el tamaño en bytes de lo que queremos recuperar. Y así, dependiendo del tipo de dato a recuperar. En nuestro caso recuperamos los datos del tipo REG_SZ.

Con esto cerramos la parte del registro de Windows, como pueden darse cuenta no es tan complicado utilizar el registro, y es bueno que se acostumbre a utilizarlo, para cuando quieran que su programa almacene información para algún uso especifico.

De todas, maneras vamos a ver ahora la segunda forma estructurada para almacenar información y me refiero a los archivos INI.

ARCHIVOS INI

Antes de ir al proyecto vamos a estudiar como esta estructurado un archivo *.ini, si abren cualquier archivo verán básicamente 2 cosas, 

(TITULO(
Data = Valor

Data2 = Valor

(TITULO2(
Data = Valor

Data2 = Valor

Un archivo INI esta formado por secciones y cada sección, contiene sus valores, el nombre de los valores o data puede repetirse en diferentes secciones. Inclusive para recuperar algún valor siempre será necesario saber la sección donde el valor esta ubicado. Cabe mencionar que lo que llamamos “data” en la documentación oficial lo pueden ver como “key”.

Ya visto un poco la estructura del archivo INI, vamos al proyecto. El cual necesitara de 7 botones. Luego pegan este código en el formulario:

Private Declare Function WritePrivateProfileSection Lib "kernel32" Alias "WritePrivateProfileSectionA" (ByVal lpAppName As String, ByVal lpString As String, ByVal lpFileName As String) As Long

Private Declare Function WritePrivateProfileString Lib "kernel32" Alias "WritePrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As Any, ByVal lpString As Any, ByVal lpFileName As String) As Long

Private Declare Function WriteProfileSection Lib "kernel32" Alias "WriteProfileSectionA" (ByVal lpAppName As String, ByVal lpString As String) As Long

Private Declare Function WriteProfileString Lib "kernel32" Alias "WriteProfileStringA" (ByVal lpszSection As String, ByVal lpszKeyName As String, ByVal lpszString As String) As Long

Private Declare Function GetPrivateProfileInt Lib "kernel32" Alias "GetPrivateProfileIntA" (ByVal lpApplicationName As String, ByVal lpKeyName As String, ByVal nDefault As Long, ByVal lpFileName As String) As Long

Private Declare Function GetPrivateProfileSection Lib "kernel32" Alias "GetPrivateProfileSectionA" (ByVal lpAppName As String, ByVal lpReturnedString As String, ByVal nSize As Long, ByVal lpFileName As String) As Long

Private Declare Function GetPrivateProfileSectionNames Lib "kernel32.dll" Alias "GetPrivateProfileSectionNamesA" (ByVal lpszReturnBuffer As String, ByVal nSize As Long, ByVal lpFileName As String) As Long

Private Declare Function GetPrivateProfileString Lib "kernel32" Alias "GetPrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As Any, ByVal lpDefault As String, ByVal lpReturnedString As String, ByVal nSize As Long, ByVal lpFileName As String) As Long

Private Declare Function GetProfileInt Lib "kernel32" Alias "GetProfileIntA" (ByVal lpAppName As String, ByVal lpKeyName As String, ByVal nDefault As Long) As Long

Private Declare Function GetProfileSection Lib "kernel32" Alias "GetProfileSectionA" (ByVal lpAppName As String, ByVal lpReturnedString As String, ByVal nSize As Long) As Long

Private Declare Function GetProfileString Lib "kernel32" Alias "GetProfileStringA" (ByVal lpAppName As String, ByVal lpKeyName As String, ByVal lpDefault As String, ByVal lpReturnedString As String, ByVal nSize As Long) As Long

Private Sub Command1_Click()

    WritePrivateProfileSection "Seccion 1", "", "c:\ejemplo.ini"

    WritePrivateProfileSection "Seccion 2", "", "c:\ejemplo.ini"

    WritePrivateProfileSection "Seccion 3", "", "c:\ejemplo.ini"

End Sub

Private Sub Command2_Click()

    WritePrivateProfileString "Seccion 1", "Valor1", "texto", "c:\ejemplo.ini"

    WritePrivateProfileString "Seccion 1", "Valor2", "54654", "c:\ejemplo.ini"

    WritePrivateProfileString "Seccion 1", "Valor3", "Prueba", "c:\ejemplo.ini"

    WritePrivateProfileString "Seccion 2", "Valor1", "4412", "c:\ejemplo.ini"

    WritePrivateProfileString "Seccion 2", "Valor2", "41", "c:\ejemplo.ini"

    WritePrivateProfileString "Seccion 2", "Valor3", "Prueba", "c:\ejemplo.ini"

End Sub

Private Sub Command3_Click()

    Dim valor As Integer

    valor = GetPrivateProfileInt("Seccion 2", "Valor1", 0, "c:\ejemplo.ini")

    MsgBox valor

End Sub

Private Sub Command5_Click()

    Dim valor As String

    Dim caract As Integer

    valor = Space(255)

    caract = GetPrivateProfileString("Seccion 1", "Valor3", "Defecto", valor, 255, "c:\ejemplo.ini")

    MsgBox Left$(valor, caract)

End Sub

Private Sub Command6_Click()

    Dim valor As String

    Dim caract As Integer

    valor = Space(1000)

    caract = GetPrivateProfileSection("Seccion 1", valor, 1000, "c:\ejemplo.ini")

    Debug.Print valor

End Sub

Private Sub Command7_Click()

    Dim valor As String

    Dim caract As Integer

    valor = Space(1000)

    caract = GetPrivateProfileSectionNames(valor, 1000, "c:\ejemplo.ini")

    Debug.Print valor

End Sub

Private Sub Form_Load()

   Me.Height = 4290

   Me.Width = 2805

      Command7.Caption = "GetPrivateProfileSectionNames"

      Command7.Height = 510

      Command7.Left = 45

      Command7.TabIndex = 5

      Command7.Top = 3285

      Command7.Width = 2500

      Command6.Caption = "GetPrivateProfileSection"

      Command6.Height = 510

      Command6.Left = 45

      Command6.TabIndex = 4

      Command6.Top = 2655

      Command6.Width = 2500

      Command5.Caption = "GetPrivateProfileString"

      Command5.Height = 510

      Command5.Left = 45

      Command5.Top = 2025

      Command5.Width = 2500

      Command3.Caption = "GetPrivateProfileInt"

      Command3.Height = 510

      Command3.Left = 45

      Command3.Top = 1395

      Command3.Width = 2500

      Command2.Caption = "WritePrivateProfileString"

      Command2.Height = 510

      Command2.Left = 45

      Command2.Top = 765

      Command2.Width = 2500

      Command1.Caption = "WritePrivateProfileSection"

      Command1.Height = 510

      Command1.Left = 45

      Command1.Top = 135

      Command1.Width = 2500

End Sub
Pueden ir presionando los botones uno por uno, y medida que lo presionan vean el archivo “c:\ejemplo.ini”, que es el archivo que crea nuestro programa.

WritePrivateProfileSection

Api: Declare Function WritePrivateProfileSection Lib "kernel32" Alias "WritePrivateProfileSectionA" (ByVal lpAppName As String, ByVal lpString As String, ByVal lpFileName As String) As Long

Esta función nos sirve para crear una sección, y un posible “dato” o “key” asociado a dicha sección 

“lpAppName” = colocamos aquí la sección a crear.

“lpString” = posible “data” que queramos crear cuando se cree la sección.

“lpFileName” = ruta del archivo INI, si el archivo no existe la propia función lo crea.

En nuestro ejemplo lo usamos en el Command1:

    WritePrivateProfileSection "sección 1", "", "c:\ejemplo.ini"

    WritePrivateProfileSection "sección 2", "", "c:\ejemplo.ini"

    WritePrivateProfileSection "sección 3", "", "c:\ejemplo.ini"

Con esto creamos 3 secciones vacías, en caso que inicialmente una de las secciones tenga algún valor seria algo así:

    WritePrivateProfileSection "Seccion 1", "Data = valor", "c:\ejemplo.ini"

WritePrivateProfileString

Api: Declare Function WritePrivateProfileString Lib "kernel32" Alias "WritePrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As Any, ByVal lpString As Any, ByVal lpFileName As String) As Long

Esta función sirve para crear “key” o “data” del tipo string. 

“lpApplicationName” = sección en donde se creara el “key”.

“lpKeyName” = nombre del “key” o “data”.

“lpString” = valor que contendrá el “key”.

“lpFileName” = ruta del archivo INI.

Ejemplo:

WritePrivateProfileString "sección 1", "Valor1", "texto", "c:\ejemplo.ini"

Introducimos en la “sección 1” , el “Valor1 = texto”.

En caso de querer pasar un número, traten a ese número como de tipo String.

Ejemplo:

WritePrivateProfileString "sección 2", "Valor1", "4412", "c:\ejemplo.ini"

GetPrivateProfileInt

Api: Declare Function GetPrivateProfileInt Lib "kernel32" Alias "GetPrivateProfileIntA" (ByVal lpApplicationName As String, ByVal lpKeyName As String, ByVal nDefault As Long, ByVal lpFileName As String) As Long

Función utilizada para recuperar de una sección un valor de tipo entero.

“lpApplicationName” = sección en donde se encuentra el “key”.

“lpKeyName” = nombre del “key” o “data”.

“nDefault” = valor por defecto, que se emplea en caso de no existir el “key”, pasado en “lpKeyName”.

“lpFileName” = ruta del archivo INI.

Ejemplo:

Dim valor As Integer

valor = GetPrivateProfileInt("sección 2", "Valor1", 0, "c:\ejemplo.ini")

MsgBox valor

Como pueden ver el valor que retorna es el valor entero que solicitamos. “nDefault” significa que en caso de que no exista “Valor1”, la función retorna el valor por defecto que en nuestro caso es 0. Pero puede ser cualquier valor.

GetPrivateProfileString

Api: Declare Function GetPrivateProfileString Lib "kernel32" Alias "GetPrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As Any, ByVal lpDefault As String, ByVal lpReturnedString As String, ByVal nSize As Long, ByVal lpFileName As String) As Long

Función utilizada para recuperar de una sección un valor de tipo string.

“lpApplicationName” = sección en donde se encuentra el “key”.

“lpKeyName” = nombre del “key” o “data”.

“lpDefault” = valor por defecto, que se emplea en caso de no existir el “key”, pasado en “lpKeyName”.

“lpReturnesString” = variable en donde se almacenara la cadena de texto.

“nSize” = tamaño en bytes, de la cantidad de caracteres que queremos recuperar.

“lpFileName” = ruta del archivo INI.

Ejemplo:

Dim valor As String

Dim caract As Integer

valor = Space(255)

caract = GetPrivateProfileString("sección 1", "Valor3", "Defecto", valor, 255, "c:\ejemplo.ini")

MsgBox Left$(valor, caract)

En este caso la función retorna, la cantidad de caracteres que se recupero, dicha cantidad la almacena la variable “caract”.

GetPrivateProfileSection

Api: Declare Function GetPrivateProfileSection Lib "kernel32" Alias "GetPrivateProfileSectionA" (ByVal lpAppName As String, ByVal lpReturnedString As String, ByVal nSize As Long, ByVal lpFileName As String) As Long

Función que sirve para retornar todos los “data” o “key” con sus respectivos valores de una sección especifica.

“lpApplicationName” = sección en donde se encuentra el “key”.

“lpReturnesString” = variable en donde se almacenara todos los “data” con sus respectivo valor.

“nSize” = tamaño en bytes, de la cantidad de caracteres que queremos recuperar.

“lpFileName” = ruta del archivo INI.

Ejemplo:

Dim valor As String

Dim caract As Integer

valor = Space(1000)

caract = GetPrivateProfileSection("sección 1", valor, 1000, "c:\ejemplo.ini")

Debug.Print valor

Como verán en la ventana “inmediato” de VB, podemos ver todos los “data” con su respectivo valor de la “sección 1”.

GetPrivateProfileSectionNames

Api: Declare Function GetPrivateProfileSectionNames Lib "kernel32.dll" Alias "GetPrivateProfileSectionNamesA" (ByVal lpszReturnBuffer As String, ByVal nSize As Long, ByVal lpFileName As String) As Long

Función que sirve para retornar todas las secciones de un archivo INI en específico. 

“lpszReturnesBuffer” = variable en donde se almacenara todas las secciones del archivo INI.

“nSize” = tamaño en bytes, de la cantidad de caracteres que queremos recuperar.

“lpFileName” = ruta del archivo INI.

Ejemplo:

Dim valor As String

Dim caract As Integer

valor = Space(1000)

caract = GetPrivateProfileSectionNames(valor, 1000, "c:\ejemplo.ini")

Debug.Print valor

Hace lo mismo que la función anterior con la diferencia que con esta función tomamos el nombre de todas las secciones que posee el archivo INI  en cuestión.

WriteProfileSection

Api: Declare Function WriteProfileSection Lib "kernel32" Alias "WriteProfileSectionA" (ByVal lpAppName As String, ByVal lpString As String) As Long

WriteProfileString

Api: Declare Function WriteProfileString Lib "kernel32" Alias "WriteProfileStringA" (ByVal lpszSection As String, ByVal lpszKeyName As String, ByVal lpszString As String) As Long

GetProfileInt

Api: Declare Function GetProfileInt Lib "kernel32" Alias "GetProfileIntA" (ByVal lpAppName As String, ByVal lpKeyName As String, ByVal nDefault As Long) As Long

GetProfileSection

Api: Declare Function GetProfileSection Lib "kernel32" Alias "GetProfileSectionA" (ByVal lpAppName As String, ByVal lpReturnedString As String, ByVal nSize As Long) As Long

GetProfileString

Api: Declare Function GetProfileString Lib "kernel32" Alias "GetProfileStringA" (ByVal lpAppName As String, ByVal lpKeyName As String, ByVal lpDefault As String, ByVal lpReturnedString As String, ByVal nSize As Long) As Long

Estas funciones no usadas en el código, hacen exactamente lo mismos que sus homologas de arriba, pueden ver que la diferencia en el nombre es que no tienen la palabra “Private”, y que no tienen como parámetro “lpFileName”, y esto se debe a que estas funciones sirven para escribir directamente en el archivo “win.ini”.

¿Qué usar? En realidad Microsoft quiere que la gente se olvide de los archivos INI y usen el registro de Windows, pero en mi opinión todo dependerá del proyecto a realizar, es claro que el registro de Windows es mas estructurado, mas versátil, permite mas tipos de valores, etc, pero algunas ocasiones los archivos INI nos saquen de aprietos de manera rápida y sencilla, ya que como se habrán dado cuenta, manejar el registro a nivel de programación es mas complicado que manejar los archivo INI. Pero algunas veces seria “tonto” usar el registro para almacenar un solo valor. Para ello seria mas optimo los archivos INI. Pero como dije todo dependerá del proyecto a realizar.

Capítulo  XI. Un poco de comunicación

En este Capítulo  voy a dar una breve explicación de algunas técnicas para hacer que 2 o mas aplicaciones se comuniquen entre sí, pero voy escribir sobre lo mas básico, ya que estudiar las técnicas avanzadas para la comunicación de aplicaciones, requerirían de un manual totalmente independiente, por ejemplo, hablar sobre Winsock, ActiveX, COM, etc, son temas que requieren de un estudio minucioso y profundo. En este Capítulo  trabajare con cosas sencillas que nos pueden sacar de aprietos cuando no necesitamos de algo tan avanzado.

ATOMO

En la vida el átomo se podría decir que es lo mas simple y pequeño que existe, en la computadora o para Windows es algo parecido, “Atom” o átomo, es tan sencillamente una información ya sea de tipo “cadena” o “entero”, y a esta información se le asigna un numero ID único, el cual se emplea para acceder a dicha información, que es almacenada por Windows en una tabla que el propio Windows controla. Tenemos 2 tipos de átomos “global” y “local”, los átomos locales solo son visto por la misma aplicación, mientras que los globales pueden ser visto por otras aplicaciones. Dicho átomos globales tienen restricción, y es que solo se pueden añadir 37 átomos globales. Es decir, no podemos tener 38 átomos globales en nuestra tabla, tendría que eliminar 1 para poder insertar el otro.

Para este proyecto solo necesitan 8 botones y 1 etiqueta o “label”, luego pegan en el formulario este código:

Private Declare Function AddAtom Lib "kernel32" Alias "AddAtomA" (ByVal lpString As String) As Integer

Private Declare Function DeleteAtom Lib "kernel32" (ByVal nAtom As Integer) As Integer

Private Declare Function FindAtom Lib "kernel32" Alias "FindAtomA" (ByVal lpString As String) As Integer

Private Declare Function GetAtomName Lib "kernel32" Alias "GetAtomNameA" (ByVal nAtom As Integer, ByVal lpBuffer As String, ByVal nSize As Long) As Long

Private Declare Function GlobalAddAtom Lib "kernel32" Alias "GlobalAddAtomA" (ByVal lpString As String) As Integer

Private Declare Function GlobalDeleteAtom Lib "kernel32" (ByVal nAtom As Integer) As Integer

Private Declare Function GlobalFindAtom Lib "kernel32" Alias "GlobalFindAtomA" (ByVal lpString As String) As Integer

Private Declare Function GlobalGetAtomName Lib "kernel32" Alias "GlobalGetAtomNameA" (ByVal nAtom As Integer, ByVal lpBuffer As String, ByVal nSize As Long) As Long

Private Declare Function InitAtomTable Lib "kernel32" (ByVal nSize As Long) As Long

Dim atomo As Long

Private Sub Command1_Click()

    atomo = AddAtom("Hola")

    Label1 = "ID del nuevo atomo creado = " & atomo

End Sub

Private Sub Command2_Click()

    Label1 = vbNullString

    atomo = FindAtom("Hola")

    Label1 = "ID del atomo encontrado = " & atomo

End Sub

Private Sub Command3_Click()

    Label1 = vbNullString

    Dim borrado As Long

    borrado = DeleteAtom(atomo)

    If borrado = 0 Then

        Label1 = "Atomo borrado"

    Else

        Label1 = "Error al borrar"

    End If

End Sub

Private Sub Command4_Click()

    Label1 = vbNullString

    Dim stratomo As String

    stratomo = Space(255)

    GetAtomName atomo, stratomo, 255

    Label1 = Trim(stratomo)

End Sub

Private Sub Command5_Click()

    Label1 = vbNullString

    Dim stratomo As String

    stratomo = Space(255)

    GlobalGetAtomName atomo, stratomo, 255

    Label1 = Trim(stratomo)

End Sub

Private Sub Command6_Click()

    Label1 = vbNullString

    Dim borrado As Long

    borrado = GlobalDeleteAtom(atomo)

    If borrado = 0 Then

        Label1 = "Atomo borrado"

    Else

        Label1 = "Error al borrar"

    End If

End Sub

Private Sub Command7_Click()

    Label1 = vbNullString

    atomo = GlobalFindAtom("HolaGlobal")

    Label1 = "ID del atomo encontrado = " & atomo

End Sub

Private Sub Command8_Click()

    atomo = GlobalAddAtom("HolaGlobal")

    Label1 = "ID del nuevo atomo creado = " & atomo

End Sub

Private Sub Form_Load()

    Me.Width = 4875

    Me.Height = 3750

    Label1.Left = 90

    Label1.Top = 2655

    Label1.Height = 555

    Label1.Width = 6360

    Label1 = vbNullString

    Command1.Height = 510

    Command1.Left = 45

    Command1.Top = 90

    Command1.Width = 1815

    Command1.Caption = "Añadir Atomo Local"

    Command2.Height = 510

    Command2.Left = 45

    Command2.Top = 675

    Command2.Width = 1815

    Command2.Caption = "Buscar Atomo Local"

    Command3.Height = 510

    Command3.Left = 45

    Command3.Top = 1260

    Command3.Width = 1815

    Command3.Caption = "Borrar Atomo Local"

    Command4.Height = 510

    Command4.Left = 45

    Command4.Top = 1845

    Command4.Width = 1815

    Command4.Caption = "Obtener Atomo Local"

    Command5.Height = 510

    Command5.Left = 2160

    Command5.Top = 1845

    Command5.Width = 2355

    Command5.Caption = "Obtener Atomo Global Local"

    Command6.Height = 510

    Command6.Left = 2160

    Command6.Top = 1260

    Command6.Width = 2355

    Command6.Caption = "Borrar Atomo Global Local"

    Command7.Height = 510

    Command7.Left = 2160

    Command7.Top = 675

    Command7.Width = 2355

    Command7.Caption = "Buscar Atomo Global Local"

    Command8.Height = 510

    Command8.Left = 2160

    Command8.Top = 90

    Command8.Width = 2355

    Command8.Caption = "Añadir Atomo Global Local"

End Sub

Private Sub Form_Unload(Cancel As Integer)

    DeleteAtom atomo

    GlobalDeleteAtom atomo

End Sub
Con este proyecto van a generar un ejecutable, que lo guardan en su disco duro, luego ejecutan su aplicación 2 veces, es decir, deben de haber dos aplicaciones del mismo proyecto arrancando al mismo tiempo. Y presionan los botones de la siguiente manera, imaginemos que ustedes tienen las dos aplicaciones abiertas, a una de ellas la llamaremos App1 y a la otra App2. A la que ustedes le hayan asignado App1 presionan el botón “Añadir Atomo Global Local”,  en la aplicación App2, presionarán el botón “Buscar Atomo Global Local”, y luego presionan “Obtener Atomo Global Local”, y verán que la etiqueta muestra “HolaGlobal", es decir, indirectamente App1 se esta comunicando con App2. Ya que una variable creada con App1, esta siendo vista por App2.

Los 4 botones hacen lo mismo, pero a nivel “local” cuando hablo de “local” me refiero a la aplicación, es decir, el programa que fue ejecutado. No vean “local” como a la propia computadora. ¿Cómo se logra todo esto?. Voy a explicar con detalle las funciones que se usan para los átomos “globales” y la explicación es análoga para los átomos “locales”.

GlobalAddAtom

Api: Declare Function GlobalAddAtom Lib "kernel32" Alias "GlobalAddAtomA" (ByVal lpString As String) As Integer

Función que se encarga de añadir un átomo global a la tabla de átomos del sistema, le pasamos como único parámetro la cadena “string” que queremos almacenar. Si la función es un éxito retorna un valor distinto de 0, que es el ID que le asigna Windows a dicho átomo.

Ej:

    atomo = GlobalAddAtom("HolaGlobal")    

    Label1 = "ID del nuevo atomo creado = " & atomo

GlobalFindAtom

Api: Declare Function GlobalFindAtom Lib "kernel32" Alias "GlobalFindAtomA" (ByVal lpString As String) As Integer

Esta función se encarga de buscar un átomo, en la lista de átomos, como único parámetro le pasamos la cadena que identifica al átomo, en caso de encontrarlo, la función retorna un valor distinto de 0 el cual es el ID que Windows le asigno en su creación. 

Ej:

    Label1 = vbNullString  

    atomo = GlobalFindAtom("HolaGlobal")    

    Label1 = "ID del atomo encontrado = " & atomo

GlobalGetAtomName

Api: Declare Function GlobalGetAtomName Lib "kernel32" Alias "GlobalGetAtomNameA" (ByVal nAtom As Integer, ByVal lpBuffer As String, ByVal nSize As Long) As Long

Función encargada de tomar la cadena que se almaceno en un átomo, se le pasa como primer parámetro el ID del átomo, como segundo parámetro se le pasa la variable que almacenara dicho valor y como tercer parámetro le pasamos la cantidad de bytes que queremos recuperar.

Ej:

    Label1 = vbNullString

    Dim stratomo As String

    stratomo = Space(255)    

    GlobalGetAtomName atomo, stratomo, 255    

    Label1 = Trim(stratomo)

GlobalDeleteAtom

Api: Declare Function GlobalDeleteAtom Lib "kernel32" (ByVal nAtom As Integer) As Integer

Función encargada de borrar un átomo de la tabla de átomos, y se le pasa como único parámetro el ID del átomo creado. La función retorna 0 si el borrado fue exitoso, en caso contrario devuelve un valor distinto de 0.

Ej:

    Label1 = vbNullString    

    Dim borrado As Long    

    borrado = DeleteAtom(atomo)    

    If borrado = 0 Then

        Label1 = "Atomo borrado"

    Else

        Label1 = "Error al borrar"

    End If

Como pueden ver es muy sencillo trabajar con los átomos, esto se puede ver como la forma mas simple de comunicar a dos aplicaciones. Y creo que esa es la razón de por que Windows en la documentación oficial lo llamo “Atom”.

Podrían preguntarse ¿Cómo puedo comunicar 2 aplicaciones si la aplicación App2 tiene que conocer el string o cadena a buscar?. Buena pregunta, pero depende el como veamos nosotros la palabra comunicación, con átomos ustedes pueden ver lo siguiente, imaginen 2 aplicaciones, cuando arranca  Aplicacion1, dicha aplicación quiere saber si Aplicacion2 se esta ejecutando, seria fácil resolver el problema usando átomos, ¿Por qué?. Aplicacion1 tendría un bucle o Timer, ejecutando la búsqueda del átomo que será un fracaso mientras Aplicacion2 no haya sido abierta, cuando Aplicación2 sea abierta, esta registra el átomo que Aplicacion1 esta buscando, y luego cuando Aplicacion1 ejecute el timer y encuentra el átomo en ese momento Aplicacion1 sabe que Aplicacion2 esta activa. ¿Eso no es comunicación?. Para mi lo es, simple pero lo es. 

Las siguientes funciones son totalmente análogas a las funciones explicadas con la diferencia que se usan para átomos locales:

AddAtom

Api: Declare Function AddAtom Lib "kernel32" Alias "AddAtomA" (ByVal lpString As String) As Integer

DeleteAtom

Api: Declare Function DeleteAtom Lib "kernel32" (ByVal nAtom As Integer) As Integer

FindAtom

Api: Declare Function FindAtom Lib "kernel32" Alias "FindAtomA" (ByVal lpString As String) As Integer

GetAtomName

Api: Declare Function GetAtomName Lib "kernel32" Alias "GetAtomNameA" (ByVal nAtom As Integer, ByVal lpBuffer As String, ByVal nSize As Long) As Long

InitAtomTable

Api: Declare Function InitAtomTable Lib "kernel32" (ByVal nSize As Long) As Long

Esta función le asigna un tamaño en bytes a la tabla de átomos local, NOTA: no funciona para la tabla global, es decir, ya mencione que solo 37 átomos globales pueden ser almacenados, pero a nivel local no hay restricción alguna, y con esta función se puede incrementar o disminuir el tamaño en bytes de dicha tabla de átomos local.

PORTAPAPELES

Todos nosotros hemos usado el “copiar”, “pegar” y “cortar”, inclusive ha sido algo muy práctico, la pregunta es ¿Adonde va esa información?, dicha información se va para el “portapapeles” véanlo como una caja donde pueden meter de ¡TODO!, imágenes, texto, paginas web, iconos, etc. ¿Dicho portapapeles puede comunicar a dos aplicaciones?. Por supuesto que si!, ustedes pueden estar en el bloc de notas y el texto que este ahí, pegarlo en un documento en Word, indirectamente están comunicando a dos aplicaciones. Pero existe una gran diferencia, y aunque no tiene por que ser estrictamente de esa manera, la diferencia es que el portapapeles lo controla el usuario ¿Cómo es eso?. El usuario es quien decide que copia en dicho portapapeles. ¿Eso quiere decir, que no puede usar 2 aplicaciones que se transfieran imágenes entre si usando el portapapeles?. Teóricamente y en la practica si lo puedes hacer, pero no es recomendable, por el simple hecho que el usuario pueda cambiar la información cuando quiera.

Es decir, puedes tener dos aplicaciones intercambiando imágenes, y resulta que el usuario no sabe que dicha aplicaciones usan el portapapeles para intercambiar dichas imágenes, y ahora imaginen que el usuario copia un texto cualquiera, las aplicaciones pueden fallar, ya que una aplicación espera de la otra una imagen, pero al copiar el texto, dicha imagen se borra, es decir, el portapapeles es algo muy volátil, ya que es el usuario quien manipula sin darse cuenta el portapapeles. Lo que quiero que vean es que el portapapeles deben de usarlo interactuando con el usuario.

Para este proyecto se necesitan los siguientes controles: 5 botones, 1 etiqueta o label, 1 picture, 2 textbox y un control CommonDialog. Luego en el formulario coloquen este código:

Private Const CF_BITMAP = 2

Private Const CF_DIB = 8

Private Const CF_DIBV5 = 17

Private Const CF_DIF = 5

Private Const CF_DSPBITMAP = &H82

Private Const CF_DSPENHMETAFILE = &H8E

Private Const CF_DSPMETAFILEPICT = &H83

Private Const CF_DSPTEXT = &H81

Private Const CF_ENHMETAFILE = 14

Private Const CF_GDIOBJFIRST = &H300

Private Const CF_GDIOBJLAST = &H3FF

Private Const CF_HDROP = 15

Private Const CF_LOCALE = 16

Private Const CF_METAFILEPICT = 3

Private Const CF_OEMTEXT = 7

Private Const CF_OWNERDISPLAY = &H80

Private Const CF_PALETTE = 9

Private Const CF_PENDATA = 10

Private Const CF_PRIVATEFIRST = &H200

Private Const CF_PRIVATELAST = &H2FF

Private Const CF_RIFF = 11

Private Const CF_SYLK = 4

Private Const CF_TEXT = 1

Private Const CF_WAVE = 12

Private Const CF_TIFF = 6

Private Const CF_UNICODETEXT = 13

Private Const WM_COPY = &H301

Private Const WM_CUT = &H300

Private Const WM_PASTE = &H302

Const LR_LOADFROMFILE = &H10

Const IMAGE_BITMAP = 0

Const IMAGE_ICON = 1

Const IMAGE_CURSOR = 2

Const IMAGE_ENHMETAFILE = 3

Private Declare Function LoadImage Lib "user32" Alias "LoadImageA" (ByVal hInst As Long, ByVal lpsz As String, ByVal dwImageType As Long, ByVal dwDesiredWidth As Long, ByVal dwDesiredHeight As Long, ByVal dwFlags As Long) As Long

Private Declare Function OpenClipboard Lib "user32" (ByVal hwnd As Long) As Long

Private Declare Function GetOpenClipboardWindow Lib "user32" () As Long

Private Declare Function CloseClipboard Lib "user32" () As Long

Private Declare Function EmptyClipboard Lib "user32" () As Long

Private Declare Function GetClipboardOwner Lib "user32" () As Long

Private Declare Function SetClipboardData Lib "user32" (ByVal wFormat As Long, ByVal hMem As Long) As Long

Private Declare Function EnumClipboardFormats Lib "user32" (ByVal wFormat As Long) As Long

Private Declare Function GetClipboardData Lib "user32" (ByVal wFormat As Long) As Long

Private Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Private Declare Function IsClipboardFormatAvailable Lib "user32" (ByVal wFormat As Long) As Long

Private Declare Function CreateCompatibleDC Lib "gdi32" (ByVal hdc As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Declare Function BitBlt Lib "gdi32" (ByVal hDestDC As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal dwRop As Long) As Long

Private Declare Function DeleteDC Lib "gdi32" (ByVal hdc As Long) As Long

Private Declare Function lstrlen Lib "kernel32" Alias "lstrlenA" (ByVal lpString As Long) As Long

Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" (pDst As Any, pSrc As Long, ByVal ByteLen As Long)

Private Sub Command1_Click()

    Dim hBitmap As Long

    hBitmap = LoadImage(App.hInstance, Text1.Text, IMAGE_BITMAP, 0, 0, LR_LOADFROMFILE)

    If hBitmap = 0 Then

        MsgBox "Error al cargar la imagen"

        Exit Sub

    End If

    OpenClipboard Me.hwnd

    EmptyClipboard

    r = SetClipboardData(CF_BITMAP, hBitmap)

    CloseClipboard

End Sub

Private Sub Command2_Click()

    CommonDialog1.Filter = "Archivos Grafico *.bmp | *.bmp"

    CommonDialog1.ShowOpen

    Text1.Text = CommonDialog1.FileName

End Sub

Private Sub Command3_Click()

    Text2.SelStart = 0

    Text2.SelLength = Len(Text2.Text)

    SendMessage Text2.hwnd, WM_COPY, 0, 0

    Text2.SelStart = 1

End Sub

Private Sub Command4_Click()

        Dim dc As Long

        Dim bitmap As Long

        If IsClipboardFormatAvailable(CF_BITMAP) = 0 Then

            MsgBox "No hay una imagen en el portapapeles"

        Else

            OpenClipboard Me.hwnd

            bitmap = GetClipboardData(CF_BITMAP)

            dc = CreateCompatibleDC(Picture1.hdc)

            SelectObject dc, bitmap

            BitBlt Picture1.hdc, 0, 0, Picture1.Width \ 15, Picture1.Height \ 15, dc, 0, 0, vbSrcCopy

            DeleteDC dc

            CloseClipboard

        End If

End Sub

Private Sub Command5_Click()

    Dim objeto As Long, longitud As Long, buffer As String

    If IsClipboardFormatAvailable(CF_TEXT) = 0 Then

            MsgBox "No hay texto en el portapapeles"

    Else

        OpenClipboard Me.hwnd

        objeto = GetClipboardData(CF_TEXT)

        If objeto <> 0 Then

            longitud = lstrlen(objeto)

            buffer = Space$(longitud)

            CopyMemory ByVal buffer, ByVal objeto, longitud

            Label1 = "Se ha pegado = " & buffer

        End If

        CloseClipboard

    End If

End Sub

Private Sub Form_Load()

    Command1.Height = 375

    Command1.Left = 90

    Command1.Top = 540

    Command1.Width = 1905

    Command1.Caption = "Copiar Imagen"

    Command2.Height = 285

    Command2.Left = 4725

    Command2.Top = 135

    Command2.Width = 285

    Command2.Caption = "..."

    Command3.Height = 375

    Command3.Left = 90

    Command3.Top = 1485

    Command3.Width = 1905

    Command3.Caption = "Copiar texto"

    Command4.Height = 375

    Command4.Left = 2745

    Command4.Top = 540

    Command4.Width = 1905

    Command4.Caption = "Recuperar Imagen"

    Command5.Height = 375

    Command5.Left = 2745

    Command5.Top = 1485

    Command5.Width = 1905

    Command5.Caption = "Recuperar texto"

    Label1.Height = 465

    Label1.Left = 90

    Label1.Top = 2025

    Label1.Width = 4650

    Picture1.Height = 2580

    Picture1.Left = 5040

    Picture1.Top = 135

    Picture1.Width = 2670

    Text1.Height = 285

    Text1.Left = 90

    Text1.Top = 135

    Text1.Width = 4560

    Text2.Height = 285

    Text2.Left = 90

    Text2.Top = 1080

    Text2.Width = 4605

    Me.Width = 7860

    Me.Height = 3195

End Sub
En el programa pueden hacer lo siguiente: copiar una imagen o copiar un texto,  nuestro ejemplo se limita solo a esto, pero se pueden colocar en el portapapeles otros formatos de información. Para copiar una imagen lo que tienen que hacer es ubicar la imagen y luego presionar el botón “Copiar Imagen”, luego le dan al botón “Recuperar imagen” y dicha imagen es mostrada en el Picture1.

Para copiar el texto es el mismo procedimiento, colocan el texto a copiar luego presionan “Copiar Texto” y luego “Recuperar Texto” para mostrar el texto copiado en el portapapeles.

OpenClipboard

Api: Declare Function OpenClipboard Lib "user32" (ByVal hwnd As Long) As Long

Función encargada de abrir el portapapeles (Clipboard), el único parámetro que se le pasa es el “hwnd” de la aplicación que tendrá el control de modificar y recuperar la información del portapapeles.

EmptyClipboard

Api: Declare Function EmptyClipboard Lib "user32" () As Long

Función que se encarga de limpiar el portapapeles, esta función debe ser ejecutada inmediatamente después de OpenClipboard, para así darle el control del portapapeles al “hwdn” de la aplicación que ejecuto OpenClipboard.

CloseClipboard

Api: Declare Function CloseClipboard Lib "user32" () As Long

Función que cierra el portapapeles.

NOTA: Antes de explicar las funciones que vienen a continuación, vamos tocar el siguiente tema que son los “formatos”, que quiero decir con esto, todos nos daremos cuenta que prácticamente podemos pegar de todo en el portapapeles, podemos pegar una imagen, un texto, una pagina web, etc. Pero esto tiene un fondo mas abstracto, por ejemplo, si nosotros tenemos Corel Draw o Photoshop, y estamos manipulando una imagen, dicha imagen en el programa es un objeto, con propiedades propias de corel draw o photoshop, es decir, tamaño, nombre, efectos, etc. Pero la pregunta seria ¿Por qué si yo copio en corel y pego en el mismo corel heredo todas esas propiedades, y en paint no?. simple respuesta Paint no esta adaptado a ese formato de imagen, pero fíjense que se pega la imagen, pero la imagen como información en bytes nada mas. Todos los extras que corel coloca no lo pega.

También por ejemplo, cuando copiamos una pagina WEB, y la pegan en Word, es muy distinto al pegado en el bloc de notas. ¿Por qué en un programa copia un formato y en el otro no lo pega igual?. Depende de cómo fue diseñado el programa. Todo en el portapapeles se identifican con un formato. Esto le sirve a los programas de  indicativo de que es lo que hay adentro. Por ejemplo en TextBox no pueden pegar una imagen pero si un texto. ¿Cómo sabe la maquina que hay en el portapapeles?. Eso la sabe gracias a los formatos. Existen formatos personalizados, que pueden usar los programas para copiar y pegar sus propios objetos. Como existen formatos universales, es decir, que pueden ser utilizados por cualquier programa. Como siempre los formatos son constantes y su uso lo veremos a continuación, pero primero muestro algunas de las constantes del sistema que se usan para especificar los formatos:

Private Const CF_BITMAP = 2 ‘ Se utiliza para los bitmap.

Private Const CF_DIB = 8 ‘ Es un objeto que presenta la estructura BITMAPINFO

Private Const CF_DIBV5 = 17 ‘ Objeto que presenta la estructura BITMAP5VHEADER solo usado en NT.

Private Const CF_DIF = 5 ‘  Software Arts' Data Interchange Format.

Private Const CF_DSPBITMAP = &H82 ‘ Se refiere a un bitmap pero asociado a un formato privado.

Private Const CF_DSPENHMETAFILE = &H8E ‘ Se refiere a un METAFILE pero asociado a un formato privado.

Private Const CF_DSPMETAFILEPICT = &H83 ‘ Metafile-picture, asociado a un formato privado.

Private Const CF_DSPTEXT = &H81 ‘ Texto asociado a un formato privado.

Private Const CF_ENHMETAFILE = 14 ´  Nos referimos a un archivo METAFILE.

Private Const CF_HDROP = 15 ‘  Nos referimos a un manejador del tipo HDROP que identifica a una lista de archivo.

Private Const CF_METAFILEPICT = 3 ‘ Objeto METAFILEPICT.

Private Const CF_OEMTEXT = 7 ‘  Formato de texto que contiene caracteres OEM.

Private Const CF_PALETTE = 9 ‘  Se especifica un manejador de alguna paleta de colores.

Private Const CF_RIFF = 11 ‘ Representa información de audio, que es mucho mas compleja que CF_WAVE.

Private Const CF_SYLK = 4 ‘Microsoft Symbolic Link (SYLK) format
Private Const CF_TEXT = 1 ‘  Texto en formato ANSI.

Private Const CF_WAVE = 12 ‘  Representa información de audio.

Private Const CF_TIFF = 6 ‘  Tagged-image file format.
Private Const CF_UNICODETEXT = 13’  Texto en formato unicode.

En esta manual no vamos a trabajar con formatos privados, pero pueden dirigirse a la documentación oficial para averiguar mas sobre el asunto en caso de  que lo necesiten,  aquí trabajaremos con los que a mi parecer son los mas usados que son el texto y la imagen.

SetClipboardData

Api: Declare Function SetClipboardData Lib "user32" (ByVal wFormat As Long, ByVal hMem As Long) As Long

Esta función nos sirve para pegar una información en el portapapeles, como primer parámetro se le pasa el formato de la información que vamos a pegar. Como segundo parámetro le pasamos la información en memoria del objeto en cuestión.

Veamos el ejemplo:

Dim hBitmap As Long

hBitmap = LoadImage(App.hInstance, Text1.Text, IMAGE_BITMAP, 0, 0, LR_LOADFROMFILE) ‘ Cargo el objeto.

If hBitmap = 0 Then

        

MsgBox "Error al cargar la imagen"

Exit Sub

End If    

OpenClipboard Me.hwnd ‘  Abro el portapapeles

EmptyClipboard’ Vacio el portapapeles

r = SetClipboardData(CF_BITMAP, hBitmap)’  Pego la imagen especificando CF_BITMAP. 

CloseClipboard’ Cierro el portapapeles.

IsClipboardFormatAvailable

Api: Declare Function IsClipboardFormatAvailable Lib "user32" (ByVal wFormat As Long) As Long

Función que determina si el portapapeles contiene una información con un formato en específico. Como único parámetro le pasamos el formato a consultar.

Si el formato esta disponible entonces retorna un valor distinto de 0, en caso de haber otro formato retornara 0.

GetClipboardData

Api: Declare Function GetClipboardData Lib "user32" (ByVal wFormat As Long) As Long

Función que sirve para recuperar la información o el contenido almacenado en el portapapeles. Le pasamos como único parámetro el formato de la información.

Ejemplo:

        Dim dc As Long

        Dim bitmap As Long

        If IsClipboardFormatAvailable(CF_BITMAP) = 0 Then ‘ Verificamos si la información que hay en el portapapeles es una imagen.

            MsgBox "No hay una imagen en el portapapeles"

        Else

            OpenClipboard Me.hwnd

            bitmap = GetClipboardData(CF_BITMAP) ‘ Almacenamos la “data” o información en la variable “bitmap”

            dc = CreateCompatibleDC(Picture1.hdc)

            SelectObject dc, bitmap

            BitBlt Picture1.hdc, 0, 0, Picture1.Width \ 15, Picture1.Height \ 15, dc, 0, 0, vbSrcCopy

            DeleteDC dc

            CloseClipboard

        End If

Dense cuenta que para copiar el texto que se encontraba en el textbox use un mensaje que se puede utilizar en algunos controles, como lo es WM_COPY.

Ejemplo:

SendMessage Text2.hwnd, WM_COPY, 0, 0 ‘ Copia el texto que se encuentra en el textbox en el portapapeles.

GetOpenClipboardWindow

Api:  Declare Function GetOpenClipboardWindow Lib "user32" () As Long

Función que retorna el “handle” de la ventana que posee el control actual del portapapeles. Recuerden que cuando usan OpenClipboard, la aplicación que la invoca es como el usuario actual del portapapeles, con esta función sabemos que aplicación controla el portapapeles en un determinado momento.

Como pueden ver, esto es algo que parece muy sencillo pero puede resultar algo muy complejo, aquí presento un vistazo general al portapapeles, pero el funcionamiento completo del portapapeles puede ser un trabajo arduo y duro. En la documentación oficial podrán encontrar mucha mas información, acerca de Formatos Múltiples, y el uso de formatos privados.

MAILSLOT

Hemos visto maneras y formas de comunicarnos en una maquina local. Ciertamente la curiosidad estará en si es posible comunicarme con otras maquinas en red. En realidad hay muchas maneras de comunicarse con maquinas en una red, lo mas optimo seria usar WinSock, dicha tecnología provee de funciones y herramientas que optimizan y aseguran el envió de información, esto implica que aprender dicha tecnología no es tan fácil y hay que estar familiarizado con las terminologías de redes.

En este manual no trabajaremos con WinSock ya que esa tecnología requieren de un manual aparte por lo largo y lo complejo que puede resultar. Habrá momentos donde necesitemos comunicar 2 computadoras para hacer algún proceso sencillo o simple. Windows provee de varias  herramientas de las cuales estudiaremos el MailSlot. Esta técnica nos provee de herramientas relativamente simples para la comunicación entre dos maquinas. 

Un MailSlot es como un archivo en memoria en donde se almacenan los mensajes que la computadora recibe, inclusive van a ver en el código que se manipula como un archivo, por eso es importante que hayan leído el Capítulo  que habla sobre archivos. Como un archivo en memoria podemos leer y escribir en él, el proceso de comunicación lo hace Windows a través de las funciones Api’s. Así que empecemos con los proyectos, para este ejemplo se necesitan 2 proyectos de VB, uno que va ser el cliente y el otro que será el servidor, a nivel de diseño son iguales y a nivel de código son parecidos excepto por unas 3 líneas.

Para cada proyecto es necesario una etiqueta o “label”, un botón, un TextBox, y un listbox.

En el proyecto cliente colocan este código:

Option Explicit

Private Const FILE_ATTRIBUTE_NORMAL = &H80

Private Const GMEM_FIXED = &H0

Private Const GMEM_MOVEABLE = &H2

Private Const GMEM_DDESHARE = &H2000

Private Const GMEM_SHARE = &H2000

Private Const GMEM_DISCARDABLE = &H100

Private Const GMEM_NOT_BANKED = &H1000

Private Const GMEM_LOWER = GMEM_NOT_BANKED

Private Const GMEM_NOCOMPACT = &H10

Private Const GMEM_NODISCARD = &H20

Private Const GMEM_NOTIFY = &H4000

Private Const GMEM_ZEROINIT = &H40

Private Declare Function CreateFile Lib "kernel32" Alias "CreateFileA" (ByVal lpFileName As String, ByVal dwDesiredAccess As Long, ByVal dwShareMode As Long, lpSecurityAttributes As Long, ByVal dwCreationDisposition As Long, ByVal dwFlagsAndAttributes As Long, ByVal hTemplateFile As Long) As Long

Private Declare Function GlobalAlloc Lib "kernel32" (ByVal wFlags As Long, ByVal dwBytes As Long) As Long

Private Declare Function GlobalReAlloc Lib "kernel32" (ByVal hMem As Long, ByVal dwBytes As Long, ByVal wFlags As Long) As Long

Private Declare Function GlobalFree Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Function GlobalUnlock Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" (pDst As Any, pSrc As Any, ByVal ByteLen As Long)

Private Declare Function lstrlen Lib "kernel32" Alias "lstrlenA" (ByVal lpString As String) As Long

Private Declare Function ReadFile Lib "kernel32" (ByVal hfile As Long, lpBuffer As Any, ByVal nNumberOfBytesToRead As Long, lpNumberOfBytesRead As Long, lpOverlapped As Long) As Long

Private Declare Function WriteFile Lib "kernel32" (ByVal hfile As Long, lpBuffer As Any, ByVal nNumberOfBytesToWrite As Long, lpNumberOfBytesWritten As Long, lpOverlapped As Long) As Long

Private Declare Function CloseHandle Lib "kernel32" (ByVal hObject As Long) As Long

'Private Declare Function CreateMailslot Lib "kernel32" Alias "CreateMailslotA" (ByVal lpName As String, ByVal nMaxMessageSize As Long, ByVal lReadTimeout As Long, lpSecurityAttributes As SECURITY_ATTRIBUTES) As Long

Private Declare Function CreateMailslot Lib "kernel32" Alias "CreateMailslotA" (ByVal lpName As String, ByVal nMaxMessageSize As Long, ByVal lReadTimeout As Long, lpSecurityAttributes As Long) As Long

Private Declare Function GetMailslotInfo Lib "kernel32" (ByVal hMailslot As Long, lpMaxMessageSize As Long, lpNextSize As Long, lpMessageCount As Long, lpReadTimeout As Long) As Long

Private Declare Function SetMailslotInfo Lib "kernel32" (ByVal hMailslot As Long, ByVal lReadTimeout As Long) As Long

Private Const INVALID_HANDLE_VALUE = -1

Private Const OPEN_EXISTING = 3

Private Const FILE_SHARE_READ = &H1

Private Const GENERIC_WRITE = &H40000000

Private Const MAILSLOT_NO_MESSAGE = (-1)

Private Const MAILSLOT_WAIT_FOREVER = (-1)

Dim hmail As Long

Dim s1 As String

Dim bBytes() As Byte

Dim hfile As Long

Dim espacio As Long

Dim tamaño As Long

Dim cadena As String

Private Sub Command1_Click()

    hfile = CreateFile("\\.\mailslot\servidor", GENERIC_WRITE, FILE_SHARE_READ, ByVal 0&, OPEN_EXISTING, FILE_ATTRIBUTE_NORMAL, 0)

    If hfile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If

    Dim espacio1 As Long

    s1 = Text1.Text

    espacio1 = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1))

    CopyMemory ByVal espacio1, ByVal s1, Len(s1)

    Dim ret As Long

    WriteFile hfile, ByVal espacio1, Len(s1), ret, ByVal 0&

    'Debug.Print "ret write = " & ret

    GlobalFree espacio1

    CloseHandle hfile

End Sub

Private Sub Form_Load()

    hmail = CreateMailslot("\\.\mailslot\cliente", 0, MAILSLOT_WAIT_FOREVER, ByVal 0&)

    Command1.Height = 420

    Command1.Left = 4815

    Command1.Top = 450

    Command1.Width = 1230

    Command1.Caption = "Enviar"

    Label1.Height = 510

    Label1.Left = 135

    Label1.Top = 450

    Label1.Width = 3345

    Label1.Caption = vbNullString

    List1.Height = 840

    List1.Left = 135

    List1.Top = 945

    List1.Width = 5910

    Text1.Height = 330

    Text1.Left = 135

    Text1.Top = 90

    Text1.Width = 5910

    Text1.Text = vbNullString

    Me.Caption = "cliente"

    Me.Height = 2280

    Me.Width = 6270

End Sub

Private Sub Form_Unload(Cancel As Integer)

    CloseHandle hmail

End Sub

Private Sub Timer1_Timer()

    Dim hResult As Long

    Dim nsize As Long

    Dim mcount As Long

    Dim s2 As String

    Dim ret As Long

    hResult = GetMailslotInfo(hmail, ByVal 0&, nsize, mcount, ByVal 0&)

    If hResult = 0 Then

        Label1 = "MALO"

        Exit Sub

    End If

    If nsize = MAILSLOT_NO_MESSAGE Then

        Label1 = "No hay mensajes"

        Exit Sub

    End If

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, nsize)

    ReadFile hmail, ByVal espacio, nsize, ret, ByVal 0

    s2 = String$(ret, 0)

    CopyMemory ByVal s2, ByVal espacio, ret

    Label1 = "Hay un mensaje"

    List1.AddItem s2

    GlobalFree espacio

End Sub

En el proyecto servidor colocan este código:

Option Explicit

Private Const FILE_ATTRIBUTE_NORMAL = &H80

Private Const GMEM_FIXED = &H0

Private Const GMEM_MOVEABLE = &H2

Private Const GMEM_DDESHARE = &H2000

Private Const GMEM_SHARE = &H2000

Private Const GMEM_DISCARDABLE = &H100

Private Const GMEM_NOT_BANKED = &H1000

Private Const GMEM_LOWER = GMEM_NOT_BANKED

Private Const GMEM_NOCOMPACT = &H10

Private Const GMEM_NODISCARD = &H20

Private Const GMEM_NOTIFY = &H4000

Private Const GMEM_ZEROINIT = &H40

Private Declare Function CreateFile Lib "kernel32" Alias "CreateFileA" (ByVal lpFileName As String, ByVal dwDesiredAccess As Long, ByVal dwShareMode As Long, lpSecurityAttributes As Long, ByVal dwCreationDisposition As Long, ByVal dwFlagsAndAttributes As Long, ByVal hTemplateFile As Long) As Long

Private Declare Function GlobalAlloc Lib "kernel32" (ByVal wFlags As Long, ByVal dwBytes As Long) As Long

Private Declare Function GlobalReAlloc Lib "kernel32" (ByVal hMem As Long, ByVal dwBytes As Long, ByVal wFlags As Long) As Long

Private Declare Function GlobalFree Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Function GlobalUnlock Lib "kernel32" (ByVal hMem As Long) As Long

Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" (pDst As Any, pSrc As Any, ByVal ByteLen As Long)

Private Declare Function lstrlen Lib "kernel32" Alias "lstrlenA" (ByVal lpString As String) As Long

Private Declare Function ReadFile Lib "kernel32" (ByVal hfile As Long, lpBuffer As Any, ByVal nNumberOfBytesToRead As Long, lpNumberOfBytesRead As Long, lpOverlapped As Long) As Long

Private Declare Function WriteFile Lib "kernel32" (ByVal hfile As Long, lpBuffer As Any, ByVal nNumberOfBytesToWrite As Long, lpNumberOfBytesWritten As Long, lpOverlapped As Long) As Long

Private Declare Function CloseHandle Lib "kernel32" (ByVal hObject As Long) As Long

'Private Declare Function CreateMailslot Lib "kernel32" Alias "CreateMailslotA" (ByVal lpName As String, ByVal nMaxMessageSize As Long, ByVal lReadTimeout As Long, lpSecurityAttributes As SECURITY_ATTRIBUTES) As Long

Private Declare Function CreateMailslot Lib "kernel32" Alias "CreateMailslotA" (ByVal lpName As String, ByVal nMaxMessageSize As Long, ByVal lReadTimeout As Long, lpSecurityAttributes As Long) As Long

Private Declare Function GetMailslotInfo Lib "kernel32" (ByVal hMailslot As Long, lpMaxMessageSize As Long, lpNextSize As Long, lpMessageCount As Long, lpReadTimeout As Long) As Long

Private Declare Function SetMailslotInfo Lib "kernel32" (ByVal hMailslot As Long, ByVal lReadTimeout As Long) As Long

Private Const INVALID_HANDLE_VALUE = -1

Private Const OPEN_EXISTING = 3

Private Const FILE_SHARE_READ = &H1

Private Const GENERIC_WRITE = &H40000000

Private Const MAILSLOT_NO_MESSAGE = (-1)

Private Const MAILSLOT_WAIT_FOREVER = (-1)

Dim hmail As Long

Dim s1 As String

Dim bBytes() As Byte

Dim hfile As Long

Dim espacio As Long

Dim tamaño As Long

Dim cadena As String

Private Sub Command1_Click()

    hfile = CreateFile("\\.\mailslot\cliente", GENERIC_WRITE, FILE_SHARE_READ, ByVal 0&, OPEN_EXISTING, FILE_ATTRIBUTE_NORMAL, 0)

    If hfile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If

    Dim espacio1 As Long

    s1 = Text1.Text

    espacio1 = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, Len(s1))

    CopyMemory ByVal espacio1, ByVal s1, Len(s1)

    Dim ret As Long

    WriteFile hfile, ByVal espacio1, Len(s1), ret, ByVal 0&

    'Debug.Print "ret write = " & ret

    GlobalFree espacio1

    CloseHandle hfile

End Sub

Private Sub Form_Load()

    hmail = CreateMailslot("\\.\mailslot\servidor", 0, MAILSLOT_WAIT_FOREVER, ByVal 0&)

    Command1.Height = 420

    Command1.Left = 4815

    Command1.Top = 450

    Command1.Width = 1230

    Command1.Caption = "Enviar"

    Label1.Height = 510

    Label1.Left = 135

    Label1.Top = 450

    Label1.Width = 3345

    Label1.Caption = vbNullString

    List1.Height = 840

    List1.Left = 135

    List1.Top = 945

    List1.Width = 5910

    Text1.Height = 330

    Text1.Left = 135

    Text1.Top = 90

    Text1.Width = 5910

    Text1.Text = vbNullString

    Me.Caption = "servidor"

    Me.Height = 2280

    Me.Width = 6270

End Sub

Private Sub Form_Unload(Cancel As Integer)

    CloseHandle hmail

End Sub

Private Sub Timer1_Timer()

    Dim hResult As Long

    Dim nsize As Long

    Dim mcount As Long

    Dim s2 As String

    Dim ret As Long

    hResult = GetMailslotInfo(hmail, ByVal 0&, nsize, mcount, ByVal 0&)

    If hResult = 0 Then

        Label1 = "MALO"

        Exit Sub

    End If

    If nsize = MAILSLOT_NO_MESSAGE Then

        Label1 = "No hay mensajes"

        Exit Sub

    End If

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, nsize)

    ReadFile hmail, ByVal espacio, nsize, ret, ByVal 0

    s2 = String$(ret, 0)

    CopyMemory ByVal s2, ByVal espacio, ret

    Label1 = "Hay un mensaje"

    List1.AddItem s2

    GlobalFree espacio

End Sub

Para arrancar estos programas tienen dos opciones, si tienen los proyectos abiertos pues ejecútenlos, o pueden compilar y ejecutar los respectivos .exe, haga como lo hagan abran los dos programas, en el textbox coloquen un texto cualquiera y luego denle al botón “Enviar” si estaban en el cliente el mensaje será recibido por el Servidor y viceversa, dicho mensaje lo coloco en el ListBox. Ustedes pueden preguntar ¿Pero estos programas no están red, sino local?. Tienen toda la razón, pero el programa fue diseñado para que funcione de esa manera, cuando explique las funciones involucradas en el proyecto, explicare como hacerlo compatible a una red. Y quedará en ustedes hacer la prueba.

Como mencione el MailSlot es como un archivo, y como archivo hay que crearlo:

CreateMailslot

Api: Declare Function CreateMailslot Lib "kernel32" Alias "CreateMailslotA" (ByVal lpName As String, ByVal nMaxMessageSize As Long, ByVal lReadTimeout As Long, lpSecurityAttributes As Long) As Long

Api: Declare Function CreateMailslot Lib "kernel32" Alias "CreateMailslotA" (ByVal lpName As String, ByVal nMaxMessageSize As Long, ByVal lReadTimeout As Long, lpSecurityAttributes As SECURITY_ATTRIBUTES) As Long

Como recordarna de otras funciones mostradas mas arriba, en nuestro proyecto usamos la primera donde no es necesario pasarle la estructura SECURITY_ATTRIBUTES, ya que dicha estructura esta diseñada para usarla en NT.

Esta función se utiliza para crear un MailSlot. Como primer parametro “lpName” le pasamos la ubicación o direccion “virtual” de nuestro MailSlot. Pueden escribirse de la manera siguiente:

"\\.\mailslot\cliente" ‘ Asi creas un mailslot local.

"\\NombreComputadora\mailslot\cliente" ‘ Asi creas un mailslot remoto.

Puedes usar una jerarquia, por ejemplo:

"\\NombreComputadora\mailslot\carpeta1\carpeta2\cliente" 
"\\.\mailslot\carpeta1\carpeta2\cliente" 
Como segundo parámetro “nMaxMessageSize”, le pasamos el tamaño máximo en bytes, de los mensajes que se abran de almacenar en el MailSlot. Si especificamos 0, estamos diciendo que puede aceptar cualquier tamaño. Como tercer parámetro “lReadTimeout” especificamos el tiempo de espera de un mensaje, especificado en milisegundos. Al colocar MAILSLOT_WAIT_FOREVER, le especificamos que espere por siempre, hasta que sea cerrado el MailSlot. Y el ultimo parámetro no es utilizado y se le pasa 0.

Ejemplo:

hmail = CreateMailslot("\\.\mailslot\cliente", 0, MAILSLOT_WAIT_FOREVER, ByVal 0&) 

Esta línea ubicada en Form_Load, crea un MailSlot, que acepta cualquier mensaje de cualquier tamaño, y espera por siempre por dicho mensajes.

Luego para cerrar un MailSlot usamos la función CloseHandle:

CloseHandle hmail

Es muy importante que cierren los objetos que crean, recuerden que estos objetos no los controla VB por lo que si cierran el programa sin borrar los objetos ellos quedaran en memoria.

¿Como enviar informacion?
Enviar información es exactamente igual que almacenar información en un archivo:

    hfile = CreateFile("\\.\mailslot\servidor", GENERIC_WRITE, FILE_SHARE_READ, ByVal 0&, OPEN_EXISTING, FILE_ATTRIBUTE_NORMAL, 0) ‘  Creo un archivo, en el nombre coloco la direccion del MailSlot al cual le quiero escribir o enviar un mensaje.

    If hfile = INVALID_HANDLE_VALUE Then

        MsgBox "error al cargar el archivo"

    End If


.


.


.

    WriteFile hfile, ByVal espacio1, Len(s1), ret, ByVal 0& ‘ Escribo en el MailSlot, aqui Windows se encarga de enviar a través de la red nuestra información.

Pueden ver que el envió es muy sencillo y no se necesita de ninguna función especial, sino las vistas cuando explicamos la sección de archivos.

¿Como recibimos informacion?
Para esto si tenemos que usar una función nueva:

GetMailslotInfo

Api: Declare Function GetMailslotInfo Lib "kernel32" (ByVal hMailslot As Long, lpMaxMessageSize As Long, lpNextSize As Long, lpMessageCount As Long, lpReadTimeout As Long) As Long

Esta función es utilizada para poder recuperar los mensajes localizados en un MailSlot. Como primer parámetro le pasamos el “handle” del Mailslot del cual queremos extraer dichos mensajes, como segundo parámetro tenemos el máximo tamaño permitido para almacenar los mensajes, como tercer parámetro “lpNextSize” tenemos el tamaño en bytes del próximo mensaje a recuperar. Y como último parámetro tenemos la cantidad de tiempo en milisegundos, que la operación de lectura puede esperar por un mensaje. 

Vayamos al ejemplo:

    hResult = GetMailslotInfo(hmail, ByVal 0&, nsize, mcount, ByVal 0&) ‘ Nos preparamos a recibir el próximo mensaje.

    If hResult = 0 Then

        Label1 = "MALO"

        Exit Sub

    End If

    If nsize = MAILSLOT_NO_MESSAGE Then ‘ Si nsize retorna esta constante, implica que no hay mensajes en la cola

        Label1 = "No hay mensajes"

        Exit Sub

    End If

    espacio = GlobalAlloc(GMEM_FIXED Or GMEM_ZEROINIT, nsize)’ Preparamos la memoria en donde almacenaremos la información.

    ReadFile hmail, ByVal espacio, nsize, ret, ByVal 0’ Leemos la informacion como si de un archivo se tratara.

    s2 = String$(ret, 0)

    CopyMemory ByVal s2, ByVal espacio, ret ‘ Preparamos la información para ser mostrada.

    Label1 = "Hay un mensaje"

    List1.AddItem s2 ‘ Mostramos el contenido.

Todo esto lo pongo en un timer, con el sentido de que la aplicación puede recibir algún mensaje en cualquier momento.

Como pueden ver, es relativamente sencillo trabajar con MailSlot, ya que se trata prácticamente de conocer mas las funciones de archivos que del propio MailSlot, si se dan cuenta solo usamos 2 funciones nuevas. De todas maneras en la documentación MSDN pueden encontrar información mas detallada de esta tecnología.

Capítulo  XII. Funciones Api’s Informativas 

En este Capítulo  tratare de algunas API’S usadas para retornar información del sistema y además, veremos como apagamos y reiniciamos nuestra computadora y por último estudiaremos las funciones utilizadas para cambiar la resolución del sistema, funciones muy utilizadas en caso de desarrollar un juego de computadora.

UN POCO DE INFORMACIÓN
Para este proyecto necesitaremos de un ComboBox y un ListBox. Luego pegan este código:

Private Const SM_ARRANGE = 56

Private Const SM_CLEANBOOT = 67

Private Const SM_CMETRICS = 44

Private Const SM_CMONITORS = 80

Private Const SM_CMOUSEBUTTONS = 43

Private Const SM_CXBORDER = 5

Private Const SM_CXCURSOR = 13

Private Const SM_CXDLGFRAME = 7

Private Const SM_CXDOUBLECLK = 36

Private Const SM_CXDRAG = 68

Private Const SM_CXEDGE = 45

Private Const SM_CXFIXEDFRAME = SM_CXDLGFRAME

Private Const SM_CXFRAME = 32

Private Const SM_CXFULLSCREEN = 16

Private Const SM_CXHSCROLL = 21

Private Const SM_CXHTHUMB = 10

Private Const SM_CXICON = 11

Private Const SM_CXICONSPACING = 38

Private Const SM_CXMAXIMIZED = 61

Private Const SM_CXMAXTRACK = 59

Private Const SM_CXMENUCHECK = 71

Private Const SM_CXMENUSIZE = 54

Private Const SM_CXMIN = 28

Private Const SM_CXMINIMIZED = 57

Private Const SM_CXMINSPACING = 47

Private Const SM_CXMINTRACK = 34

Private Const SM_CXSCREEN = 0

Private Const SM_CXSIZE = 30

Private Const SM_CXSIZEFRAME = SM_CXFRAME

Private Const SM_CXSMICON = 49

Private Const SM_CXSMSIZE = 52

Private Const SM_CXVIRTUALSCREEN = 78

Private Const SM_CXVSCROLL = 2

Private Const SM_CYBORDER = 6

Private Const SM_CYCAPTION = 4

Private Const SM_CYCURSOR = 14

Private Const SM_CYDLGFRAME = 8

Private Const SM_CYDOUBLECLK = 37

Private Const SM_CYDRAG = 69

Private Const SM_CYEDGE = 46

Private Const SM_CYFIXEDFRAME = SM_CYDLGFRAME

Private Const SM_CYFRAME = 33

Private Const SM_CYFULLSCREEN = 17

Private Const SM_CYHSCROLL = 3

Private Const SM_CYICON = 12

Private Const SM_CYICONSPACING = 39

Private Const SM_CYMAXIMIZED = 62

Private Const SM_CYMAXTRACK = 60

Private Const SM_CYMENU = 15

Private Const SM_CYMENUCHECK = 72

Private Const SM_CYMIN = 29

Private Const SM_CYMENUSIZE = 55

Private Const SM_CYMINIMIZED = 58

Private Const SM_CYMINSPACING = 48

Private Const SM_CYMINTRACK = 35

Private Const SM_CYSCREEN = 1

Private Const SM_CYSIZE = 31

Private Const SM_CYSIZEFRAME = SM_CYFRAME

Private Const SM_CYSMCAPTION = 51

Private Const SM_CYSMICON = 50

Private Const SM_CYSMSIZE = 53

Private Const SM_CYVIRTUALSCREEN = 79

Private Const SM_CYVTHUMB = 9

Private Const SM_CYVSCROLL = 20

Private Const SM_DBCSENABLED = 42

Private Const SM_DEBUG = 22

Private Const ARW_BOTTOMLEFT = &H0&

Private Const ARW_BOTTOMRIGHT = &H1&

Private Const ARW_HIDE = &H8&

Private Const ARW_TOPLEFT = &H2&

Private Const ARW_TOPRIGHT = &H3&

Private Const VER_PLATFORM_WIN32_NT = 2

Private Const VER_PLATFORM_WIN32s = 0

Private Const VER_PLATFORM_WIN32_WINDOWS = 1

Private Const MAX_COMPUTERNAME_LENGTH = 31

Private Const MAX_PATH = 260

Private Type SYSTEM_INFO

    dwOemID As Long

    dwPageSize As Long

    lpMinimumApplicationAddress As Long

    lpMaximumApplicationAddress As Long

    dwActiveProcessorMask As Long

    dwNumberOrfProcessors As Long

    dwProcessorType As Long

    dwAllocationGranularity As Long

    dwReserved As Long

End Type

Private Const PROCESSOR_INTEL_386 = 386

Private Const PROCESSOR_INTEL_486 = 486

Private Const PROCESSOR_MIPS_R4000 = 4000

Private Const PROCESSOR_INTEL_PENTIUM = 586

Private Const PROCESSOR_ALPHA_21064 = 21064

Private Const UNLEN = 256

Private Type OSVERSIONINFO

    dwOSVersionInfoSize As Long

    dwMajorVersion As Long

    dwMinorVersion As Long

    dwBuildNumber As Long

    dwPlatformId As Long

    szCSDVersion As String * 128

End Type

Private Declare Function GetVersionEx Lib "kernel32" Alias "GetVersionExA" (lpVersionInformation As OSVERSIONINFO) As Long

Private Declare Function GetUserName Lib "advapi32.dll" Alias "GetUserNameA" (ByVal lpBuffer As String, nSize As Long) As Long

Private Declare Sub GetSystemInfo Lib "kernel32" (lpSystemInfo As SYSTEM_INFO)

Private Declare Function GetComputerName Lib "kernel32" Alias "GetComputerNameA" (ByVal lpBuffer As String, nSize As Long) As Long

Private Declare Function GetKeyboardType Lib "user32" (ByVal nTypeFlag As Long) As Long

Private Declare Function GetSystemDirectory Lib "kernel32" Alias "GetSystemDirectoryA" (ByVal lpBuffer As String, ByVal nSize As Long) As Long

Private Declare Function GetSystemMetrics Lib "user32" (ByVal nIndex As Long) As Long

Private Declare Function GetWindowsDirectory Lib "kernel32" Alias "GetWindowsDirectoryA" (ByVal lpBuffer As String, ByVal nSize As Long) As Long

Private Sub Form_Load()

    Combo1.AddItem "GetComputerName"

    Combo1.AddItem "GetKeyboardType"

    Combo1.AddItem "GetSystemDirectory"

    Combo1.AddItem "GetSystemInfo"

    Combo1.AddItem "GetSystemMetrics"

    Combo1.AddItem "GetUserName"

    Combo1.AddItem "GetVersionEx"

    Combo1.AddItem "GetWindowsDirectory"

End Sub

Private Sub Combo1_Click()

    List1.Clear

    Select Case Combo1.ListIndex

    Case 0

        Call NombreComputadora

    Case 1

        Call TipoTeclado

    Case 2

        Call WindowSystem

    Case 3

        Call InformacionSistema

    Case 4

        Call SystemMetrics

    Case 5

        Call Usuario

    Case 6

        Call OsVersion

    Case 7

        Call DirectorioWindow

    End Select

End Sub

Sub DirectorioWindow()

    Dim nombre As String

    Dim res As Long

    nombre = Space(MAX_PATH)

    res = GetWindowsDirectory(nombre, MAX_PATH)

    List1.AddItem "Windows  = " & Left(nombre, res)

End Sub

Sub OsVersion()

    Dim osv As OSVERSIONINFO

    osv.dwOSVersionInfoSize = Len(osv)

    GetVersionEx osv

    List1.AddItem "Major Version = " & osv.dwMajorVersion

    List1.AddItem "Menor Version = " & osv.dwMinorVersion

    List1.AddItem "Version = " & osv.dwMajorVersion & "." & osv.dwMinorVersion

    Select Case osv.dwPlatformId

    Case VER_PLATFORM_WIN32s

        List1.AddItem "Win32 Windows 3.1"

    Case VER_PLATFORM_WIN32_WINDOWS

        List1.AddItem "Win32 Windows 95, 98, Me."

    Case VER_PLATFORM_WIN32_NT

        List1.AddItem "Win32 Windows NT"

    End Select

    List1.AddItem osv.szCSDVersion

End Sub

Sub Usuario()

    Dim nombre As String

    Dim res As Long

    nombre = Space(255)

    res = 255

    GetUserName nombre, res

    List1.AddItem "Usuario = " & Left(nombre, res)

End Sub

Sub SystemMetrics()

    Dim res As Long

    res = GetSystemMetrics(SM_ARRANGE)

    Select Case res

    Case ARW_BOTTOMLEFT

        List1.AddItem "ARW_BOTTOMLEFT"

    Case ARW_BOTTOMRIGHT

        List1.AddItem "ARW_BOTTOMRIGHT"

    Case ARW_HIDE

        List1.AddItem "ARW_HIDE"

    Case ARW_TOPLEFT

        List1.AddItem "ARW_TOPLEFT"

    Case ARW_TOPRIGHT

        List1.AddItem "ARW_TOPRIGHT"

    End Select

    res = GetSystemMetrics(SM_CLEANBOOT)

    Select Case res

    Case 0

        List1.AddItem "El sistema arranco normal"

    Case 1

        List1.AddItem "El sistema arranco en modo seguro"

    Case 2

        List1.AddItem "El sistema arranco en modo seguro con red"

    End Select

    res = GetSystemMetrics(SM_CMONITORS)

    List1.AddItem "Monitores = " & res

    res = GetSystemMetrics(SM_CMOUSEBUTTONS)

    List1.AddItem "Botones del mouse = " & res

    res = GetSystemMetrics(SM_CXSCREEN)

    List1.AddItem "Resolucion Ancho = " & res

    res = GetSystemMetrics(SM_CYSCREEN)

    List1.AddItem "Resolucion Alto = " & res

    res = GetSystemMetrics(SM_CXICON)

    List1.AddItem "Ancho por defecto de un icono = " & res

    res = GetSystemMetrics(SM_CYICON)

    List1.AddItem "Alto por defecto de un icono = " & res

End Sub

Sub InformacionSistema()

    Dim sysinf As SYSTEM_INFO

    GetSystemInfo sysinf

    List1.AddItem "Page Size = " & sysinf.dwPageSize

    List1.AddItem "Minimum Application Address = " & sysinf.lpMinimumApplicationAddress

    List1.AddItem "Maximum Application Address = " & sysinf.lpMaximumApplicationAddress

    List1.AddItem "Numero de procesadores = " & sysinf.dwNumberOrfProcessors

    Select Case sysinf.dwProcessorType

    Case PROCESSOR_INTEL_386

        List1.AddItem "Tipo procesador = INTEL 386"

    Case PROCESSOR_INTEL_486

        List1.AddItem "Tipo procesador = INTEL 486"

    Case PROCESSOR_INTEL_PENTIUM

        List1.AddItem "Tipo procesador = INTEL PENTIUM"

    Case PROCESSOR_MIPS_R4000 ' solo para NT

        List1.AddItem "Tipo procesador = MIPS"

    Case PROCESSOR_ALPHA_21064 ' solo para NT

        List1.AddItem "Tipo procesador = MIPS"

    End Select

End Sub

Sub TipoTeclado()

    Dim res As Long

    res = GetKeyboardType(0)

    Select Case res

    Case 1

        List1.AddItem "IBM PC/XT or compatible (83-key) keyboard"

    Case 2

        List1.AddItem "Olivetti ICO (102-key) keyboard"

    Case 3

        List1.AddItem "IBM PC/AT (84-key) or similar keyboard"

    Case 4

        List1.AddItem "IBM enhanced (101- or 102-key) keyboard"

    Case 5

        List1.AddItem "Nokia 1050 and similar keyboards"

    Case 6

        List1.AddItem "Nokia 9140 and similar keyboards"

    Case 7

        List1.AddItem "Japanese keyboard"

    End Select

End Sub

Sub WindowSystem()

    Dim nombre As String

    Dim res As Long

    nombre = Space(MAX_PATH)

    res = GetSystemDirectory(nombre, MAX_PATH)

    List1.AddItem "Windows System = " & Left(nombre, res)

End Sub

Sub NombreComputadora()

    Dim nombre As String

    nombre = Space(MAX_COMPUTERNAME_LENGTH + 1)

    GetComputerName nombre, MAX_COMPUTERNAME_LENGTH + 1

    List1.AddItem "Nombre de Computadora = " & nombre

End Sub

El ejecutar dicho programa y pasear por el comboBox notarán como el LisBox se va llenando de información; a continuación estudiaremos las funciones que nos ayuda a retornar dicha información.

GetComputerName

Api: Declare Function GetComputerName Lib "kernel32" Alias "GetComputerNameA" (ByVal lpBuffer As String, nSize As Long) As Long

Función encargada de retornar el nombre de la computadora; nombre que se le fue asignado al momento de instalar Windows, aunque dicho nombre puede ser modificado por el panel de control. O una función que veremos mas adelante.

Como primer parámetro le pasamos la variable que almacenara el nombre, y como segundo parámetro le pasamos la cantidad de bytes que queremos recuperar. Como pueden ver su uso es muy sencillo.

SetComputerName

Api: Declare Function SetComputerName Lib "kernel32" Alias "SetComputerNameA" (ByVal lpComputerName As String) As Long

Función no usada en el proyecto, dicha función es la encargada de asignarle un nuevo nombre a la computadora, como pueden ver es muy simple ya que como único parámetro tenemos la variable del tipo String que contiene el nombre que le queramos asignar. Teniendo claro que el limite de caracteres o bytes viene establecido por la constante:

Private Const MAX_COMPUTERNAME_LENGTH = 31

GetKeyboardType

Api: Declare Function GetKeyboardType Lib "user32" (ByVal nTypeFlag As Long) As Long

Función encargada de retornar información acerca de la configuración del teclado. Como único parámetro le pasamos un “flag” que indica que tipo de información queremos consultar. Particularmente puede tomar 3 valores:

0 – Tipo de teclado

1 – Sub-tipo de teclado

2 – Numero de teclas “funcionales” en el teclado.

Si colocamos 0, podemos obtener estos resultados:

	Valor
	Significado

	1
	IBM PC/XT or compatible (83-key) keyboard

	2
	Olivetti "ICO" (102-key) keyboard

	3
	IBM PC/AT (84-key) or similar keyboard

	4
	IBM enhanced (101- or 102-key) keyboard

	5
	Nokia 1050 and similar keyboards

	6
	Nokia 9140 and similar keyboards

	7
	Japanese keyboard


Si colocamos 2, podemos obtener estos resultados:

	Valor
	Numero de teclas de “funcion”

	1
	10

	2
	12 (sometimes 18)

	3
	10

	4
	12

	5
	10

	6
	24

	7
	Hardware dependent and specified by the OEM


Dicho valor se toma de lo que retorna la función cuando es ejecutada. En nuestro caso veamos el ejemplo:

    res = GetKeyboardType(0)

La variable “res” contendrá el resultado que habrá de devolver la función al ser ejecutada, como el flag fue 0, “res” estará comprendido entre los valores de la primera tabla.

GetSystemDirectory

Api: Declare Function GetSystemDirectory Lib "kernel32" Alias "GetSystemDirectoryA" (ByVal lpBuffer As String, ByVal nSize As Long) As Long

Función encargada de retornar el directorio SYSTEM. Es decir, se encarga de señalar la ubicación de dicho directorio, normalmente es C:\Windows\System, o para los usuarios NT podria ser C:\Winnt\System32.

Como primer parámetro le pasamos la variable que almacenará dicho valor, como segundo parámetro le pasamos la cantidad de bytes que queremos recuperar, y la función devuelve o retorna la cantidad de bytes que realmente fueron recuperados.

En el ejemplo es fácil ver lo que hace:

    nombre = Space(MAX_PATH) ‘Configuramos nuestra variable de tipo string

    res = GetSystemDirectory(nombre, MAX_PATH) ‘Llamamos la función y la cantidad de bytes a recuperar sera el valor de la constante MAX_PATH

    List1.AddItem "Windows System = " & Left(nombre, res) ‘Tomamos solo los bytes necesarios. Y lo mostramos en el List1.

GetSystemInfo

Api: Declare Sub GetSystemInfo Lib "kernel32" (lpSystemInfo As SYSTEM_INFO)

Función que retorna información actual del sistema, dicha información se almacena en una estructura que estudiaremos a continuación:

Private Type SYSTEM_INFO

    dwOemID As Long ‘ Parámetro obsoleto, que solo se mantiene por compatibilidad.

    dwPageSize As Long ‘ Especifica el tamaño de pagina, que puede ser utilizado con la función VirtualAlloc.

    lpMinimumApplicationAddress As Long ‘ Puntero de la dirección baja en memoria accesible para aplicaciones y DLL

    lpMaximumApplicationAddress As Long ‘ Puntero de la dirección alta en memoria accesible para aplicaciones y DLL
    dwActiveProcessorMask As Long ‘ 

    dwNumberOrfProcessors As Long ‘ Número de procesadores que posee el computador.

    dwProcessorType As Long ‘ Tipo de procesador presente en el computador.

    dwAllocationGranularity As Long

    dwReserved As Long ‘ Parámetro reservado por el sistema.

End Type
En el parametro dwProcessorType, tenemos varios tipos:

Private Const PROCESSOR_INTEL_386 = 386

Private Const PROCESSOR_INTEL_486 = 486

Private Const PROCESSOR_INTEL_PENTIUM = 586

Solo para NT:

Private Const PROCESSOR_MIPS_R4000 = 4000

Private Const PROCESSOR_ALPHA_21064 = 21064

GetSystemMetrics

Api: Declare Function GetSystemMetrics Lib "user32" (ByVal nIndex As Long) As Long

Con esta función podemos retornar información métrica (es decir, ancho y alto de distinto componentes) y además algunas configuraciones del sistema. Todas las medidas se encuentran especificadas en píxeles. 

La información que queremos buscar, se le pasa como constante en el parámetro “nIndex” y lo que retorna la función es el valor que estábamos buscando. En el ejemplo no ataque todas las constantes ya que son muchas y coloque las mas importantes, de todas maneras a continuación se explica para que sirve cada constante que puede ser pasada en dicha función:

Dejo en claro que las constantes que empiezen por SM_CY, se refiere a “alto” y las que empiecen por SM_CX, se refiere a “alto”.

Private Const SM_ARRANGE = 56 ‘ Flag que sirve para conocer como el sistema ordena las ventanas minimizadas.

Private Const SM_CLEANBOOT = 67 ‘ Con este parámetro podemos saber como se inicio el sistema. Si la función retorna 0 = Booteo normal, 1 = Arranque en modo seguro, 2 = Arranque modo seguro con compatibilidad de red.

Private Const SM_CMONITORS = 80 ‘ Sirve para conocer la cantidad de monitores, conectados al sistema.

Private Const SM_CMOUSEBUTTONS = 43 ‘ Sirve para conocer la cantidad de botones presente en el mouse.

Private Const SM_CXBORDER = 5 
Private Const SM_CYBORDER = 6 ‘ Ancho y alto del borde de las ventanas.
Private Const SM_CXCURSOR = 13 
Private Const SM_CYCURSOR = 14 ‘ Ancho y alto del cursor de windows
Private Const SM_CXDLGFRAME = 7

Private Const SM_CYDLGFRAME = 8 ‘ Hace lo mismo que SM_CXFIXEDFRAME y SM_CYFIXEDFRAME
Private Const SM_CXDOUBLECLK = 36

Private Const SM_CYDOUBLECLK = 37 ‘ Ancho y alto del rectangulo localizado alrededor del primer click.
Private Const SM_CXEDGE = 45

Private Const SM_CYEDGE = 46 ‘ Ancho y alto del borde 3-D
Private Const SM_CXFIXEDFRAME = SM_CXDLGFRAME

Private Const SM_CYFIXEDFRAME = SM_CYDLGFRAME ‘ Ancho y alto del rectangulo o “frame” alrededor del perimetro de la ventana que presenta “Caption” pero no es redimensionable.
Private Const SM_CXFULLSCREEN = 16

Private Const SM_CYFULLSCREEN = 17 ‘ Ancho y alto del area cliente, establecido por una ventana maximizada.
Private Const SM_CXHSCROLL = 21

Private Const SM_CYHSCROLL = 3 ‘ Ancho de la flecha presente en un scroll horizontal, y alto del scroll horizontal.
Private Const SM_CXHTHUMB = 10 ‘ Ancho del “thumb” del scroll horizontal. Thumb es la barra por donde se desplaza el cuadrito que vemos en una barra de desplazamiento.

Private Const SM_CYVTHUMB = 9 ‘ Alto del “thumb” del scroll vertical. 
Private Const SM_CXICON = 11

Private Const SM_CYICON = 12 ‘ Ancho y alto por defecto de un icono.
Private Const SM_CXMAXIMIZED = 61

Private Const SM_CYMAXIMIZED = 62 ‘ Ancho y alto por defecto de una ventana cuando se maximiza.
Private Const SM_CXMENUCHECK = 71

Private Const SM_CYMENUCHECK = 72 ‘ Ancho y alto por defecto de la casilla de verificacion colocada en el menu. 
Private Const SM_CXMIN = 28

Private Const SM_CYMIN = 29 ‘ Ancho y alto mínimo que puede tomar una ventana.
Private Const SM_CXMINIMIZED = 57

Private Const SM_CYMINIMIZED = 58 ‘ Ancho y alto de una ventana minimizada.
Private Const SM_CXSCREEN = 0

Private Const SM_CYSCREEN = 1 ‘ Ancho y alto de la dimensión de la pantalla
Private Const SM_CXSIZE = 30

Private Const SM_CYSIZE = 31 ‘ Ancho y alto de los botones presenta en el “Caption” o barra de título.
Private Const SM_CXSMICON = 49

Private Const SM_CYSMICON = 50 ‘ Ancho y alto recomendados para los iconos pequeños
Private Const SM_CXVSCROLL = 2

Private Const SM_CYVSCROLL = 20 ‘ Alto de la flecha presente en un scroll vertical, y ancho del scroll vertical.
Private Const SM_CYCAPTION = 4 ‘ Alto de la barra de título o “Caption” de una ventana

Private Const SM_CYMENU = 15 ‘ Alto de una simple línea del menu bar.

Private Const SM_MOUSEPRESENT = 19 ‘ Indica mediante TRUE si hay un mouse instalado.

Private Const SM_MOUSEWHEELPRESENT = 75 ‘ Indica mediante TRUE o FALSE la existencia de la ruedita que poseen algunos mouse.

Estas son algunas constantes, de todas maneras revisen la documentacion oficial, ya que a medida que salen nuevas versiones de windows dicho listado de constantes puede ir aumentando.

Para la constante SM_ARRANGE, la función puede retornar algunos de los siguientes valores:

Private Const ARW_BOTTOMLEFT = &H0& ‘ Empieza en la esquina inferior izquierda de la pantalla (posicion por defecto)

Private Const ARW_BOTTOMRIGHT = &H1& ‘Empieza en la esquina inferior derecha de la pantalla.

Private Const ARW_HIDE = &H8& ‘ Esconde la ventana minimizada.

Private Const ARW_TOPLEFT = &H2& ‘  Empieza en la esquina superior izquierda de la pantalla.

Private Const ARW_TOPRIGHT = &H3& ‘  Empieza en la esquina superior derecha de la pantalla.
GetUserName

Api: Declare Function GetUserName Lib "advapi32.dll" Alias "GetUserNameA" (ByVal lpBuffer As String, nSize As Long) As Long

Esta función retorna el usuario activo. Es decir, el usuario que inicio la sesión. Como primer parámetro se le pasa la variable que almacenara el nombre del usuario, y como segundo parámetro se le pasa la cantidad de bytes a recuperar.

Por ejemplos anteriores sea intuitivo pensar que lo que retorna la función es la cantidad de bytes recuperados, pero en esta función no se trabaja de esa manera, después de haber ejecutado dicha función, la misma variables pasada en el parámetro “nSize” contendrá la cantidad de bytes recuperados.

Como se puede ver en el ejemplo del proyecto:

    nombre = Space(255)

    res = 255

    GetUserName nombre, res

    List1.AddItem "Usuario = " & Left(nombre, res)

En este caso la variable “res” contiene la cantidad de bytes recuperados.

GetVersionEx

Api: Declare Function GetVersionEx Lib "kernel32" Alias "GetVersionExA" (lpVersionInformation As OSVERSIONINFO) As Long

Función encargada de recuperar información acerca de la versión del sistema operativo. Dicha información es almacenada en una estructura, que es el único parámetro que se le pasa a la función.

Private Type OSVERSIONINFO

    dwOSVersionInfoSize As Long ‘ Tamaño de la estructura en bytes

    dwMajorVersion As Long ‘ Identifica el número “mayor” de la versión del sistema operativo.

    dwMinorVersion As Long ‘Identifica el número “menor” de la versión del sistema operativo.
    dwBuildNumber As Long 

    dwPlatformId As Long ‘ Identifica la plataforma del sistema operativo.

    szCSDVersion As String * 128 ‘ Variable que puede contener información adicional, por ejemplo, en caso de poseer NT, y tener el “Service Pack 3” instalado, en dicha constante se almacenara “Service Pack 3”, dando a entender al usuario que tiene dicha service pack instalado.

End Type
¿Menor o Mayor?. Es sencillo entender esos dos parametros, un número que representa a una versión puede estar representado de la siguiente manera: 4.15, en esta caso “mayor” será 4 y el “menor” 15. La unión de estos dos parámetros daría la versión 4.15.

El parámetro dwPlatformId, puede retornar los siguientes valores:

Private Const VER_PLATFORM_WIN32s = 0 ‘ Windows 3.1

Private Const VER_PLATFORM_WIN32_WINDOWS = 1 ‘ Windows 98, 95, ME, etc.

Private Const VER_PLATFORM_WIN32_NT = 2 ‘ Windows NT.

GetWindowsDirectory

Api: Declare Function GetWindowsDirectory Lib "kernel32" Alias "GetWindowsDirectoryA" (ByVal lpBuffer As String, ByVal nSize As Long) As Long

Función que sirve para retornar el directorio donde esta instalado el sistema operativo, y funciona exactamente igual que la función GetSystemDirectory.

APAGANDO, RESETEANDO EL SISTEMA
ExitWindowsEx

Api: Declare Function ExitWindowsEx Lib "user32" (ByVal uFlags As Long, ByVal dwReserved As Long) As Long

Función que sirve para resetear o apagar la computadora, tambien sirve para cerrar la sesión del usuario activo.

Como primer parámetro se le pasa algunas de las siguientes constantes, las cuales realizan distintas acciones:

Public Const EWX_LOGOFF = 0 ‘ Cierra todo los procesos que estan corriendo, y finaliza la sesión del usuario activo.

Public Const EWX_POWEROFF = &H8 ‘ Prepara al sistema para apagarse.

Public Const EWX_REBOOT = 2 ‘ Prepara al sistema y resetea la computadora.

Public Const EWX_SHUTDOWN = 1 ‘ Prepara al sistema para apagar la computadora, con esto me refiero a transferir información al disco duro, cerrar los programas, etc. Y luego Windows apaga la máquina.

Estos parámetros se pueden usar junto con los antes mencionados:

Public Const EWX_FORCE = 4 ‘  Se obliga a terminar todos los procesos activos, dichos procesos no mandan a las aplicaciones los mensages WM_QUERYENDSESION y WM_ENDSESION.

Estos dos últimos mensajes pueden ser capturados por los programas para realizar las operaciones pertinentes y cerrar la aplicación de manera apropiada, para evitar asi la perdida de información. Al usar EWX_FORCE se obliga a cerrar las aplicaciones, sin que estas pueden realizar las operaciones pertienentes de cierre, produciendo posibles perdidas de información.

Esta API es la utilizada en sistema Windows 95,98,ME. A continuación se mostraran funciones que solo trabajan en plataforma NT.

LockWorkStation

Api: Declare Function LockWorkStation Lib "user32.dll" () As Long

Bloquea la computadora, en la cual se ejecuta dicha función.

InitiateSystemShutdown

Api: Declare Function InitiateSystemShutdown Lib "advapi32.dll" Alias "InitiateSystemShutdownA" (ByVal lpMachineName As String, ByVal lpMessage As String, ByVal dwTimeout As Long, ByVal bForceAppsClosed As Long, ByVal bRebootAfterShutdown As Long) As Long

Esta función hace prácticamente lo mismo que ExitWindowsEx con la diferencia que se pueden apagar otras maquinas de manera remota, e incluye ciertas cosas que se mostrarán a continuación:

Como primer parámetro “lpMachineName”, le pasamos el nombre de la computadora a la cual le queremos aplicar la “acción” a ejecutar. Si se le pasa NULL se entiende que nos referimos a la maquina local. Como segundo parámetro “lpMessage”, le pasamos una cadena de tipo String, la cual se le he mostrada al usuario de dicha computadora, notificándole que dicha computadora se apagara o se reseteara, también se le puede pasar NULL si no queremos ningún mensaje. “dwTimeout” especifica el tiempo en milisegundos que durará el mensaje del parámetro “lpMessage”, por ejemplo, si pasamos 2000 que es lo equivalente a 2 segundos, entonces el mensaje se le muestra al usuario por 2 segundos y luego ejecuta la acción a seguir.

El parámetro “bForceAppsClosed”, sirve para forzar a las aplicaciones para cerrarse, es decir, si se le pasa TRUE, las aplicaciones se cierran automáticamente y sin necesidad de salvar la información. En caso de pasar FALSE se le muestran mensajes al usuario que le especifica que la aplicación va ser cerrada.

Y por último tenemos “bRebootAfterShutdown”, en caso de colocarle TRUE, la computadora se resetea, en caso de FALSE, el sistema guarda toda la información pertinente de Windows, y procede a apagar la computadora.

Queda claro, que no siempre se puede aplicar esta función ya que depende de los privilegios del usuario, ciertamente si es el administrador quien ejecuta dicha aplicación lo mas seguro es que no tenga problema, pero no cualquier usuario puede apagarla ya que dependerá si posee la permisología de la maquina para realizar dicha operación. Para ver como alterar dicha permisología, vayan directo a la documentación de MSND y vean la parte de “PRIVILEGES”

AbortSystemShutdown

Api: Declare Function AbortSystemShutdown Lib "advapi32.dll" Alias "AbortSystemShutdownA" (ByVal lpMachineName As String) As Long

Se recuerdan del mensaje que puede ser mostrado por la función anterior; mientras se muestra dicho mensaje, nosotros podemos suspender el reseteo o apagado de la maquina, y dicha suspensión se hace con esta función, la cual tiene como unico parametro el nombre del computadora a la cual se quiere abortar la ejecución de la “accion” que se quería. 

CAMBIANDO LA RESOLUCIÓN DE WINDOWS

Aquí les presento las 2 Api’s necesarias para cambiar la resolución de Windows. Para este ejemplo necesitan un formulario con un Listbox y un Botón.

Const CCDEVICENAME = 32

Const CCFORMNAME = 32

Private Type DEVMODE

    dmDeviceName As String * CCDEVICENAME

    dmSpecVersion As Integer

    dmDriverVersion As Integer

    dmSize As Integer

    dmDriverExtra As Integer

    dmFields As Long

    dmOrientation As Integer

    dmPaperSize As Integer

    dmPaperLength As Integer

    dmPaperWidth As Integer

    dmScale As Integer

    dmCopies As Integer

    dmDefaultSource As Integer

    dmPrintQuality As Integer

    dmColor As Integer

    dmDuplex As Integer

    dmYResolution As Integer

    dmTTOption As Integer

    dmCollate As Integer

    dmFormName As String * CCFORMNAME

    dmUnusedPadding As Integer

    dmBitsPerPel As Integer

    dmPelsWidth As Long

    dmPelsHeight As Long

    dmDisplayFlags As Long

    dmDisplayFrequency As Long

End Type

Const CDS_UPDATEREGISTRY = &H1

Const CDS_TEST = &H4

Const DISP_CHANGE_SUCCESSFUL = 0

Const DISP_CHANGE_RESTART = 1

Private Declare Function EnumDisplaySettings Lib "user32" Alias "EnumDisplaySettingsA" (ByVal lpszDeviceName As Long, ByVal iModeNum As Long, lpDevMode As Any) As Boolean

Private Declare Function ChangeDisplaySettings Lib "user32" Alias "ChangeDisplaySettingsA" (lpDevMode As Any, ByVal dwFlags As Long) As Long

Dim dev As DEVMODE

Dim i, resultado1, resultado2 As Long

Dim Resul As Boolean

Private Sub Command1_Click()

    dev.dmPelsWidth = Mid(List1.Text, 1, InStr(1, List1.Text, "X") - 2)

    dev.dmPelsHeight = Mid(List1.Text, InStr(1, List1.Text, "X") + 2, 3)

    dev.dmBitsPerPel = Mid(List1.Text, InStr(1, List1.Text, "X") + 5, InStr(1, List1.Text, "b") - (InStr(1, List1.Text, "X") + 5))

    resultado1 = ChangeDisplaySettings(dev, CDS_TEST)

    If resultado1 = DISP_CHANGE_RESTART Then

        MsgBox "Necesitas reiniciar la maquina"

    End If

    If resultado1 = DISP_CHANGE_SUCCESSFUL Then

        ChangeDisplaySettings dev, CDS_UPDATEREGISTRY

    End If

End Sub

Private Sub Form_Load()

    List1.Left = 135

    List1.Top = 135

    List1.Height = 2205

    List1.Width = 4110

    Command1.Left = 2835

    Command1.Top = 2430

    Command1.Height = 510

    Command1.Width = 1410

    Command1.Caption = "Cambiar Resolucion"

    Me.Height = 3435

    Me.Width = 4500

    Do

        Resul = EnumDisplaySettings(0&, i, dev)

        List1.AddItem dev.dmPelsWidth & " X " & dev.dmPelsHeight & " " & dev.dmBitsPerPel & " bits"

        i = i + 1

    Loop While Resul

End Sub

Si ejecutan el programa verán que en el ListBox se colocan todas las Resoluciones posibles que soporta la tarjeta de video que posean, seleccionen una y denle clic a “Cambiar Resolución” e inmediatamente verán como la computadora cambia de resolución. 

ENUMDISPLAYSETTINGS

Api: Declare Function EnumDisplaySettings Lib "user32" Alias "EnumDisplaySettingsA" (ByVal lpszDeviceName As Long, ByVal iModeNum As Long, lpDevMode As Any) As Boolean
Con esta API obtenemos información acerca de un dispositivo gráfico, en nuestro caso la usamos para hacer un listado de las resoluciones posibles a las cuales podemos cambiar.

Como parámetros tenemos: 

lpszDeviceName: este parámetro puede ser NULL. Si es NULL nos referimos al dispositivo gráfico usado en ese momento.

iModeNum: sirve como índice, para recuperar la información del dispositivo gráfico.

lpDevMode: Es un puntero a una estructura de tipo DEVMODE.

En el programa lo usamos de esta manera:

    Do

        Resul = EnumDisplaySettings(0&, i, dev)

        List1.AddItem dev.dmPelsWidth & " X " & dev.dmPelsHeight & " " & dev.dmBitsPerPel & " bits"

        i = i + 1

    Loop While Resul

End Sub

Como podrán observar, la estructura DEVMODE representada en la variable “dev”, cambia sus valores cada vez que entra, con este código logramos que “Mientras Resul sea true” continué mostrando información, cuando ya se haya extraído toda la información EnumDisplaySettings retorna FALSE y sale del bucle. Para moverse en los distintos tipos de información usamos la variable “i” la cual empieza en 0 y va aumentando de uno en uno.

CHANGEDISPLAYSETTINGS

Api: Declare Function ChangeDisplaySettings Lib "user32" Alias "ChangeDisplaySettingsA" (lpDevMode As Any, ByVal dwFlags As Long) As Long

Api utilizada para cambiar la configuración de un dispositivo gráfico. En nuestro caso lo usamos para cambiar la resolución del sistema operativo.

Como parámetros tenemos, “lpDevMode” el cual se le pasa una variable de tipo DEVMODE, y luego tenemos un Flag “dwFlag” el cual lo utilizamos para indicar lo que queremos hacer.

En el programa lo usamos de la manera siguiente:

    resultado1 = ChangeDisplaySettings(dev, CDS_TEST)

    If resultado1 = DISP_CHANGE_RESTART Then

        MsgBox "Necesitas reiniciar la maquina"

    End If

    If resultado1 = DISP_CHANGE_SUCCESSFUL Then

        ChangeDisplaySettings dev, CDS_UPDATEREGISTRY

    End If

Primero probamos si la resolución, a la cual estamos invocando es compatible, para ello, usamos un flag “CDS_TEST” si la función ChangeDisplaySettings retorna DISP_CHANGE_RESTART significa que para realizar el cambio de la resolución hay que reiniciar la maquina, si la función retorna DISP_CHANGE_SUCCESSFUL significa que no hay ningún problema. Ahora ¿Por qué invocar dos veces ChangeDsiplaySettings?

resultado1 = ChangeDisplaySettings(dev, CDS_TEST)

En este caso probamos si es compatible.

ChangeDisplaySettings dev, CDS_UPDATEREGISTRY

Y aquí sabiendo que ya es compatible, actualizamos esa información en el registro de Windows, y  para ello usamos el flag CDS_UPDATEREGISTRY

Aquí mostré lo básico de las dos funciones usadas para cambiar la resolución de un dispositivo gráfico. Si desean conocer, con detalle los valores que retornan, los flag, etc. Pueden consultarla en la ayuda de MSDN de Visual Studio.

Conclusión. 

Es increíble que haya terminado este manual y todavía sienta que me falta tanto por aprender, muchos dirán ¿Esto es todo? ¿Hay mas? ¿Se acabo?, ciertamente estos son mis conocimientos sobre las funciones API’S de Windows y ciertamente me falta demasiado por aprender. Como verán en el glosario solo explique 277 funciones, si tomamos en cuenta que son mas de 1000 funciones API’S podemos decir que atacamos aproximadamente un 25%.

Muchos dirán ¿Es que hay mas?, yo diría que demasiado, solo espero que las persona que hayan leído dicho manual completo, les haya servido para su trabajo diario, o les haya motivado el estudiar el interesante mundo que esconde Windows detrás de las “ventanas”, espero también que no solo se queden con los pocos conocimientos que les impartí sino que amplíen mucho mas, leyendo la ayuda de MSDN, estoy seguro que muchos de ustedes dirán: ¿Pero la ayuda de MSDN es un fastidio, ya que no se entiende?. Y yo les aseguro que si entendieron el manual, estoy seguro que cuando lean la documentación oficial tendrán todo mucho mas claro.

En esta pagina http://www16.brinkster.com/eduroam/api estará el manual en línea, y las posibles actualizaciones del mismo. De verdad, espero que les haya servido, y aquellos que hayan creados códigos o ejemplos educativos que ustedes crean que puede servir para entender alguna función que yo explique o que no fue tocada en este manual, dichos códigos serán bienvenidos para ser publicados en la pagina antes mencionada.

Así que sin mas que decir, se despide su colega.

Eduardo Roa.

Email: eduroam@cantv.net

Técnico Superior en Informática

Caracas, Venezuela.

Glosario.

Lo que quiero decir, es que ya el HDC tiene objetos predeterminados. Como una brocha estándar, un pincel, una fuente, etc. Si creamos una brocha nueva tenemos que suplantar la que tenemos por la nueva y eso es lo que hace SelectObject.





X1,Y1





X2,Y2





X3,Y3





X4,Y4





X2,Y2





X1,Y1





X3





Y3





Funciones GDI, Ejm: Rectangle(...)





Unidades Lógicas





Motor de Conversión





























Espacio Mundo





Espacio Pagina





Dispositivo





Dispositivo Físico





Hardware donde se presenta la imagen

Hola Mundo!





La distancia de esta linea es lo que hay que pasarle en el parámetro nBreakExtra	.





item.itemHeight = height_menu


item.itemWidth = width_menu


Estas lineas modifican la información del bloque de memoria de “lPAram”





Como “lParam” no es un valor como tal, sino la direccion de un bloque de memoria, funciona perfecto  debido a que “item” esta configurado como “Byref” es decir, “item”, no esta esperando una estructura como tal, esta esperando una dirección de memoria, eso si, cuyo bloque sea una estructura MEASUREITEMSTRUCT que evidentemente lo es, por que windows asegura eso.





ByVal hwnd As Long, ByVal mensaje As Long, ByVal wParam As Long, ByRef item As MEASUREITEMSTRUCT





Ubicación en memoria de “lParam”





..............





Ubicación en memoria de GestorMensajeI





En este bloque de memoria fue donde modificamos la altura y ancho.





En este bloque de memoria sigue estando la altura y ancho por defecto.





Ubicación en memoria de “item”





Ubicación en memoria de “lParam”





..............





Muevo la información 





Ubicación en memoria de “lParam”





..............





Ubicación en memoria de “item”

Área cliente





Scroll o barra de desplazamiento





Área no cliente





Menú del sistema (sysmenu)





Botones del manejo de la ventana





%s3





%s2





%s1

